
Preface

This book is a self-contained guide to problem-solving and exploration in math-
ematical physics using the powerful Maple 9.5 computer algebra system (CAS).
With a CAS one cannot only crunch numbers and plot results, but also carry out
the symbolic manipulations which form the backbone of mathematical physics.

The heart of this text consists of over 230 useful and stimulating “classic”
computer algebra worksheets or recipes, which are systematically organized to
cover the major topics presented in the standard Mathematical Physics course
offered to third or fourth year undergraduate physics and engineering students.
The emphasis here is on applications, with only a brief summary of the un-
derlying theoretical ideas being presented. The aim is to show how computer
algebra can not only implement the methods of mathematical physics quickly,
accurately, and efficiently, but can be used to explore more complex examples
which are tedious or difficult or even impossible to implement by hand.

The recipes are grouped into three sections, the introductory Appetizers
dealing with linear ordinary differential equations (ODEs), series, vectors, and
matrices. The more advanced Entrees cover linear partial differential equations
(PDEs), scalar and vector fields, complex variables, integral transforms, and
calculus of variations. Finally, in the Desserts the emphasis is on presenting
some analytic, graphical, and numerical techniques for solving nonlinear ODEs
and PDEs. The numerical methods are also applied to linear ODEs and PDEs.

No prior knowledge of Maple is assumed in this text, the relevant command
structures being introduced on a need-to-know basis. The recipes are thor-
oughly annotated and, on numerous occasions, presented in a “story” format
or in a historical context. Each recipe takes the reader from the analytic formu-
lation or statement of a representative type of mathematical physics problem to
its analytic or numerical solution and to a graphical visualization of the answer,
where relevant. The graphical representations vary from static 2-dimensional
pictures, to contour and vector field plots, to 3-dimensional graphs that can
be rotated, to animations in time. For your convenience, all 230 recipes are
included on the accompanying CD.

The range of mathematical physics problems that can be solved with the
enclosed recipes is only limited by your imagination. By altering the parameter
values, or initial conditions, or equation structure, thousands of other problems
can be easily generated and solved. “What if?” questions become answerable.
This should prove extremely useful to instructor and student alike.

Computer Algebra Recipes
for Mathematical Physics



INTRODUCTION

The purpose of computing is insight, not numbers.
R.W. Hamming, Numerical Methods for Scientists and Engineers (1973)

Science means simply the aggregate of all the recipes
that are always successful.
Paul Valéry, French poet and essayist (1871–1945)

A. Computer Algebra Systems

Computer algebra systems (CASs) are revolutionizing the way we learn and
teach those scientific subjects which make extensive use of advanced mathe-
matics. CASs not only allow us to carry out the numerical computations of
standard programming languages and to plot the results in a wide variety of
ways, but to also perform lengthy and complicated symbolic mathematical ma-
nipulations as well. The purpose of this text is to show how a CAS can be used
to tackle problem-solving and exploration of concepts and methods in mathe-
matical physics. A CAS can perform a wide variety of mathematical operations,
including

• analytic differentiation and analytic/numerical integration,

• analytic/numerical solution of ordinary/partial differential equations,

• Taylor/Laurent series expansions of functions,

• manipulation and simplification of algebraic expressions,

• analytic/numerical solution of algebraic equations,

• production of 2- and 3-dimensional vector field and contour plots,

• animation of analytic and numerical solutions.

The computer algebra worksheets, or recipes, in this book are based on the
powerful Maple 9.5 software system. Any reader desiring to use a different
release of Maple, or even an alternate CAS, should generally have little difficulty
in modifying the recipes to his or her own taste. This is because the Maple
input and output is completely annotated for each recipe and the underlying
mathematics and physics fully explained.

B. Computer Algebra Recipes

The heart of this text consists of a systematic collection of computer algebra
recipes which have been designed to illustrate the concepts and methods of
mathematical physics and to stimulate the reader’s intellect and imagination.
Associated with each recipe is an intrinsically important mathematical physics
example and, where feasible, the example is presented in a “story” format
wherein real or fictitious characters motivate or explain the recipe.

Every topic or story in the text contains the Maple code or recipe to explore
that particular topic. To make life easier for you, all recipes have been placed
on the CD-ROM enclosed within the back cover of this text. The recipes are
ordered according to the chapter number, the section number, and the subsec-
tion (story) number. For example, the recipe 01-1-2, entitled The Tale of the
Turbulent Tail, is associated with chapter 1, section 1, subsection 2. Although
the recipes can be directly accessed on the CD by clicking on the appropriate
worksheet number, it is strongly recommended that you access them through
the hyperlinked recipe index file 00recipe, which provides complete instruc-
tions. The computer code exported into the text is accompanied by detailed
explanations of the underlying mathematical physics concepts and/or methods
and what the recipe is trying to accomplish.

The recommended procedure for using this text is first to read a given
topic/story for overall understanding and enjoyment. If you are having any
difficulty in understanding a piece of the text code, then you should execute
the corresponding Maple worksheet and try variations on the code. Keep in
mind that the same objective may often be achieved by a different combination
of Maple commands than those that I chose. After reading the topic, you should
execute the worksheet (if you have not already done so) to make sure the code
works as expected. At this point feel free to explore the topic. Try rotating
any three-dimensional graphs or running any animations in the file. See what
happens when changes in the model or Maple code are made and then try to
interpret any new results. This book is intended to be open-ended and merely
serve as a guide to what is possible in mathematical physics using a CAS, the
possibilities being limited only by your own background and desires.

At the end of each chapter, Supplementary Recipes are presented in the
form of problems, their fully annotated solutions (recipes) being included on
the CD. These recipes are also hyperlinked to the recipe index file with a simple
numbering system. For example, 01-S02 is the second supplementary recipe
in Chapter 1. Supplementary recipes can be used in two different ways. They
can be regarded as problems to be solved by using the mathematical physics
concepts and computer algebra techniques presented in the main text recipes.
Your solutions can then be compared with those that I have presented. Even
if you are successful, you probably will be interested in the many little com-
puter algebra features that are introduced in my solutions. On the other hand,
these additional recipes can be regarded as still more interesting applications
of computer algebra to mathematical physics. Enjoy exploring all the recipes!

C. Maple Help
In this text, the Maple commands are introduced on a need-to-know basis. If
you wish to learn more about these commands, or about other possible com-
mands which might prove useful in solving a particular mathematical physics
problem, Maple’s Help should be consulted. The Help system allows you to
explore Maple commands and features, listed by name or subject. One can
search by topic or carry out a full text search. Both procedures are illustrated
by first using the Topic Search to find the correct form of the command for
taking a square root, and then using the Full Text Search to find the command
for analytically solving an ODE. In either case, begin by using the mouse and
clicking on Maple’s Help which opens a help window.
(a) Topic Search

• Click on Topic Search. Auto-search should then be selected.
• You wish to find the Maple command for taking the square root.

Depending on the programming language, the command could be
sqr, sqrt, root, ...In this case type sq in the Topic box. Maple will
display all the commands starting with sq. Double click on sqrt or,
alternately, single click on sqrt and then on OK. A description of
the square root command will appear on the computer screen.

(b) Full Text Search

• Click on Full Text Search.
• Type ode in the Word(s) box and click on Search.
• Double click on dsolve. A description of the dsolve command for

solving ODEs will appear along with several examples as well as
hyperlinks to related topics.

To learn more about these search methods as well as other features of Maple’s
help, open Using Help on the help page.

If on executing a Maple command, the output yields a mathematical func-
tion that is unfamiliar to you, e.g., EllipticF , you may find out what this
function is by clicking on the word to highlight it, then on Help, and finally on
Help on EllipticF. You will find that EllipticF refers to the incomplete elliptic
integral of the first kind, which is defined in the Help page. The same Help
window may also be opened by typing in a question mark followed by the word
and a semicolon, e.g., ?EllipticF;

Maple’s Help is not perfect and on occasion you might feel frustrated, but
generally it is helpful and should be consulted whenever you get stuck with
Maple syntax or are seeking just the right command to accomplish a certain
mathematical task. Maple learning and programming guides are also available
([Cha03], [MGH+03b], [MGH+03a]). Let us emphasize that in this book we will
merely scratch the surface of what can be done with the Maple symbolic com-
puting package, concentrating on those features which are relevant to tackling
mathematical physics problems.

D. Introductory Recipes

In the following chapters, recipes will be presented which correlate with the ma-
jor topics developed in standard undergraduate mathematical physics ([MW71],
[Boa83], [AW00]) texts. To give you a preliminary idea of what these recipes
will look like and to introduce some basic Maple syntax, consider the following
two kinematics examples. These introductory recipes are not on the accompa-
nying CD-ROM, so after reading the following subsections you should open up
Maple and type the recipes in and execute them.

D.1 A Dangerous Ride?

A horse is dangerous at both ends and uncomfortable in the middle.
Ian Fleming, British mystery writer, (1908–64)

The vertical displacement (in meters) of a proposed circus ride at t seconds
is given by Y =a t2 e−b t cos(c t)/(1 + d

√
t). a, b, c, and d are real constants.

(a) Determine the velocity V and acceleration A at arbitrary time t.

(b) Given a = 2 m/s2, b = 3/8 s−1, c = 10 s−1, and d = 1 s−1/2, plot V over
the time interval t = 0 to T = 20 seconds.

(c) Find the maximum V in m/s and km/h and the time at which it occurs.

(d) Plot A and V together from t = 0 to T/2 = 10 seconds and discuss the
graph. Do you think that this proposed ride is dangerous? If so adjust
the parameter values to make the ride safer.

To solve this problem, let’s first clear Maple’s internal memory of any
previously assigned values (other worksheets may be open with numerical
values given to some of the same symbols being used in the present recipe).
This is done by typing in the restart command after the opening prompt
( >) symbol, ending the command with a colon (:), and pressing Enter
(which generates a new prompt symbol) on the computer key board.

> restart:

All Maple command lines must be ended with either a colon, which suppresses
any output, or a semi-colon (;), which allows the output to be viewed.

The analytic form of the ride’s vertical coordinate Y is entered.
> Y:=a*tˆ2*exp(-b*t)*cos(c*t)/(1+d*sqrt(t));

Y :=
a t2 e(−b t) cos(c t)

1 + d
√

t
Use has been made of the assignment (:=) operator, placing Y on the left-hand
side (lhs) of the operator and the time-dependent form of Y on the right-hand
side (rhs). Assigned quantities can be mathematically manipulated. The sym-
bols *, /, +, -, and ˆ are used for multiplication, division, addition, subtraction,
and exponentiation, The Maple forms cos and exp of the cosine and exponential
commands are intuitively obvious.

Differentiating Y once with respect to t yields the velocity V ,
> V:=diff(Y,t);

V :=
2 a t e(−b t) cos(c t)

1 + d
√

t
− a t2 b e(−b t) cos(c t)

1 + d
√

t
− a t2 e(−b t) sin(c t) c

1 + d
√

t

− 1
2

a t(3/2) e(−b t) cos(c t) d

(1 + d
√

t)2
while differentiating twice yields the acceleration A.

> A:=diff(Y,t,t);

A :=
2 a e(−b t) cos(c t)

1 + d
√

t
− 4 a t b e(−b t) cos(c t)

1 + d
√

t
− 4 a t e(−b t) sin(c t) c

1 + d
√

t

− 7
4

a
√

t e(−b t) cos(c t) d

(1 + d
√

t)2
+

a t2 b2 e(−b t) cos(c t)
1 + d

√
t

+
2 a t2 b e(−b t) sin(c t) c

1 + d
√

t

+
a t(3/2) b e(−b t) cos(c t) d

(1 + d
√

t)2
− a t2 e(−b t) cos(c t) c2

1 + d
√

t

+
a t(3/2) e(−b t) sin(c t) c d

(1 + d
√

t)2
+

1
2

a t e(−b t) cos(c t) d2

(1 + d
√

t)3
The form of A would be tedious to derive by hand. With Maple, the calculation
is done quickly and without any errors. If the structure of Y is changed, the
new forms of V and A are obtained almost immediately by re-executing the
above command lines.

The given parameter values are entered. Although not necessary, I like to
leave spaces between commands on the same prompt line for easier readability.

> a:=2: b:=3/8: c:=10: d:=1: T:=20:

The velocity is plotted over the time interval t=0 to T and shown in Figure 1.
> plot(V,t=0..T);

–20

–10

0

10

20

2 4 6 8 10 12 14 16 18 20
t

Figure 1: Velocity V (vertical axis) versus time t.

By inspecting the figure, we can see that the maximum velocity occurs around
the 5 s mark and is about 24 m/s. A slightly more accurate estimate can be
obtained by placing the cursor on the top of the tallest peak in the computer
picture and clicking the mouse. The horizontal and vertical coordinates of
the cursor location are displayed in a small viewing box at the top left of
the computer screen. A much more accurate answer follows on setting the
acceleration A equal to zero and applying the floating point solve (fsolve)
command in a time range which includes the tallest peak, say t = 4 to 6 s. This
yields an answer T2 for the time to 10 digits, Maple’s default accuracy.

> T2:=fsolve(A=0,t=4..6);

T2 := 4.868771376
The maximum velocity occurs at T2 � 4.87 seconds. Then, using the eval
command to evaluate V at t = T2 ,

> Vmax:=eval(V,t=T2);

Vmax := 23.81789390
yields a maximum velocity Vmax � 23.8 m/s. The convert command with the
units option is used to convert Vmax from m/s to km/h.

> Vmax:=convert(Vmax,units,m/s,km/h);

Vmax := 85.74441804
The maximum velocity is 85 3

4 km/h, which doesn’t seem excessively high.
What about the acceleration? Let’s plot A and V together in the same figure

over the time range t = 0 to T/2 = 10 seconds. Two plot options (color and
linestyle) are introduced. A red solid line is chosen for V , a blue dashed line
for A. Note that V and A as well as the options have been entered as “Maple
lists” (the elements separated by commas and enclosed in square brackets).
Maple preserves the order and repetition of elements in a list.

> plot([V,A],t=0..T/2,color=[red,blue],linestyle=[SOLID,DASH]);

–200

–100

0

100

200

2 4 6 8 10
t

Figure 2: A (dashed curve) and V (solid) versus time t.

If you are printing pictures with multiple plots in black and white, it is partic-
ularly important to control the line style so the curves can be distinguished.

In Figure 2, we can clearly see that the acceleration is a maximum when
the velocity is zero and zero when the velocity is a maximum. The maximum
acceleration is over 200 m/s2. Since the acceleration due to gravity is about 10
m/s2, this corresponds to roughly 20 “Gees”. Do you think that such an ac-
celeration is possibly dangerous? Justify your answer. Perhaps, do an Internet
search on the effects of rapid acceleration on the human body.

Next, we look at a two-dimensional kinematics example which introduces
you to the use of a Maple library package. Library packages are very important
because they save you the effort of programming specialized plotting and math-
ematical operations. Approximately 90% of Maple’s mathematical knowledge
resides in the Maple library. Most of the recipes in this text use one or more
library packages.

D.2 The Patrol Route of Bertie Bumblebee

Belief like any other moving body follows the path of least resistance.
Samuel Butler, British author, (1835–1902)

Bertie Bumblebee, intrepid sentry for the central bee hive on the terraformed
planet Erehwon1, flies on a patrol route described t minutes after leaving the
central hive by the radial coordinate r(t) = a t2 e−b t/(1 + t2) sretem (a unit of
length on Erehwon) and the angular coordinate θ(t) = b + c t2/3 radians. a, b,
and c are real constants.
(a) Calculate Bertie’s speed V at an arbitrary time t, simplifying the result as

much as possible. Attempt to analytically determine the distance Bertie
travels in the time interval t=0 to an arbitrary time T > 0.

(b) Taking a=3, b=π/8 and c=10, determine the time it takes for Bertie to
make a complete circuit and the total distance flown.

(c) Plot Bertie’s path for the complete circuit and superimpose an animation
of his motion on this path, representing Bertie as a moving circle.

After clearing Maple’s memory with the restart command,
> restart:

Bertie’s radial and angular coordinates are entered.
> r:=a*tˆ2*exp(-b*t)/(1+tˆ2); theta:=b+c*tˆ(2/3);

r :=
a t2 e(−b t)

1 + t2
θ := b + c t(2/3)

1In 1872, the British writer Samuel Butler described a fictitious land in the utopian novel
Erewhon, the title being intended as an anagram for nowhere. In this land, the people dealt
with disease as a crime and destroyed machinery lest machines destroyed them. This would
not be the land for using computer algebra, so in the Computer Algebra Recipes series, I have
introduced a fictitious planet, Erehwon, where names are occasionally spelled backwards,
butErehwon is not backward in embracing modern technology.

Note that entering theta for the angular coordinate has produced the Greek
symbol θ in the output.

Next, Bertie’s Cartesian coordinates, X = r cos θ, Y = r sin θ, are calcu-
lated, the forms of r and θ being automatically substituted in the output.

> X:=r*cos(theta); Y:=r*sin(theta);

X :=
a t2 e(−b t) cos(b + c t(2/3))

1 + t2
Y :=

a t2 e(−b t) sin(b + c t(2/3))
1 + t2

The speed V at time t is obtained by calculating V =
√

(dX/dt)2 + (dY/dt)2.
> V:=sqrt(diff(X,t)ˆ2+diff(Y,t)ˆ2);

V := ((
2 a t e(−b t) cos(b + c t(2/3))

1 + t2
− a t2 b e(−b t) cos(b + c t(2/3))

1 + t2

−2 a t3 e(−b t) cos(b + c t(2/3))
(1 + t2)2

− 2
3

a t(5/3) e(−b t) sin(b + c t(2/3)) c

1 + t2
)2

+ (
2 a t e(−b t) sin(b + c t(2/3))

1 + t2
− a t2 b e(−b t) sin(b + c t(2/3))

1 + t2

− 2 a t3 e(−b t) sin(b + c t(2/3))
(1 + t2)2

+
2
3

a t(5/3) e(−b t) cos(b + c t(2/3)) c

1 + t2
)2)(1/2)

The output looks quite messy, so let’s simplify it, making use of the simplify
command. One of the major difficulties with simplify is that the output may
not be simplified as much as you would like or not put into a specific form
that you are trying to attain. The simplify command comes with various
optional arguments, e.g., simplify(V,symbolic) as in the following command
line, which simplifies V assuming that all the parameters are positive.

> V:=simplify(V,symbolic);

V :=
1
3
a t e(−b t)(36 − 36 b t − 36 t3 b + 9 t2 b2 + 18 t4 b2 + 9 t6 b2

+ 4 t(16/3) c2 + 8 t(10/3) c2 + 4 t(4/3) c2)(1/2)
/

(1 + t2)2

This last result is certainly simpler than the previous one, all trig terms being
eliminated. Whether it’s the simplest possible form is a matter of taste. Sim-
plifying with Maple is usually a matter of trial and error and you will see many,
many simplification examples as you progress through this book.

To determine the distance d that Bertie flies over a time interval t = 0 to
some arbitrary time T , an attempt is made to analytically evaluate the integral
d=
∫ T

0 V dt using the integration (int) command.
> d:=int(V,t=0..T);

d :=
∫ T

0

1
3
a t e(−b t)(36 − 36 b t − 36 t3 b + 9 t2 b2 + 18 t4 b2 + 9 t6 b2

+ 4 t(16/3) c2 + 8 t(10/3) c2 + 4 t(4/3) c2)(1/2)
/

(1 + t2)2dt

Maple is unable to find an analytic solution, returning the integral without
evaluating it. So, let’s enter the given parameter values, a = 3, b = π/8, and
c=10. Note that the command Pi for entering π is capitalized. Maple is case
sensitive here.

> a:=3: b:=Pi/8: c:=10:

The time T = 12.99 minutes, which is now entered, is the approximate time for
Bertie to complete one circuit. It is determined by trial and error by numerically
calculating the total distance to 4 digits, using the floating point evaluation
(evalf) command. Increasing T will not change the answer to this accuracy.

> T:=12.99; distance:=evalf(d,4);

T := 12.99 distance := 23.98
Bertie travels a total distance of about 24 sretem in one complete circuit.

To animate Bertie’s flight and superimpose the motion on a plot of the entire
route, special plots commands are required. These are contained in the plots
library package, which is now “loaded”.

> with(plots);

Warning, the name changecoords has been redefined
[animate, animate3d , ... display , ... polarplot , ... textplot3d , tubeplot ]

The with( ) command is used to load Maple library packages. Normally, I
would place a colon on the above command line to suppress the output, but
here a partial list of the large number of specialized plot commands that are
available in the plots package is shown. The commands animate, polarplot (to
plot the trajectory in polar coordinates), and display (to superimpose graphs)
in the output list will be used here. There is also a warning message that the
name changecoords has been redefined. This warning appears even if a colon
is used. If desired, warnings can be removed by using a colon and inserting
the command interface(warnlevel=0) prior to loading the library package.
From now one, I will generally artificially remove all such warnings in the text.

In the first graph, gr1, an animation of Bertie’s motion is created with
the animate command. To fit into the width of the page, the lengthy Maple
command line is broken over two text lines. Bertie’s X and Y coordinates are
entered as a Maple list. The time range is taken from t = 0 to T . I have chosen
to use 500 frames (the default is 25) to make a reasonably smooth animation.
A point style is chosen, Bertie being represented by a size 16 blue circle. A
line-ending colon is used to prevent the plotting numbers from being displayed.

> gr1:=animate([X,Y],t=0..T,frames=500,style=point,
symbol=circle,color=blue,symbolsize=16):

The polarplot command is used in gr2 to graph the entire route as a thick
(the default thickness is 0) orange line. To obtain a smooth curve, a minimum
of 500 (the default is 50) plotting points is requested.

> gr2:=polarplot([r,theta,t=0..T],numpoints=500,style=line,
color=orange,thickness=2):

The graphs are now superimposed with the display command, the axis labels

x and y being added. The double quotes denote that each enclosed item is a
“Maple string”. A string is a sequence of characters that has no value other
than itself. It cannot be assigned to, and will always evaluate to itself.

> display([gr1,gr2],labels=["x","y"]);

–1

1

y

–1 1x

Figure 3: Bertie’s patrol route while on sentry duty.

Figure 3 shows the entire path traced out by Bertie and his position (repre-
sented by the small circle) two minutes after he starts on his patrol route. The
animation can be initiated (the circle starts at the origin and moves along the
path, stopping when t=T =12.99 minutes.) by clicking on the computer plot
and then on the start arrow in the Maple tool bar at the top of the computer
screen. The animation may be made to repeat by clicking on the looped arrow
and stopped by clicking on the solid square. Other options are also available.

E. How to Use this Text

Although some of Maple’s basic syntax has been provided in these introduc-
tory recipes, it is recommended that the computer algebra novice start at the
beginning of the Appetizers, even if your mathematical physics background is
above that of the recipes presented there. It is in these early chapters that more
of the basic features of the Maple system are introduced. Further, you might be
surprised at how even initially simple problems can be made more interesting
and often much more challenging because of the fact that a computer algebra
system is being used. Whatever approach you adopt to using this book, I hope
that you savor the wide variety of mathematical physics recipes that follow.

Bon Appetit! Your computer algebra chef, Richard.



Part I

THE APPETIZERS

The last thing one discovers in composing a work
is what to put first.

Blaise Pascal, French scientist, philosopher (1623–62)

Each problem that I solved became a rule
which served afterwards to solve other problems.
René Descartes, French philosopher and mathematician (1596–1650)

Food probably has a very great influence on the
condition of men....Who knows if a well-prepared soup

was not responsible for the pneumatic pump
or a poor one for a war?

G. C. Lichtenberg, German physicist, philosopher (1742–99)



Chapter 1

Linear ODEs of Physics

In this chapter, the recipes illustrate how Maple may be used to solve and
explore some representative ordinary differential equations (ODEs) from the
world of physics. The focus is on linear ODEs (LODEs), i.e., those which
are first order or linear in the dependent variable. Although an example of a
nonlinear ODE (NLODE), i.e., one which contains one or more terms which
are not linear in the dependent variable, is presented in the second recipe, the
study of NLODEs is a much more mathematically challenging topic and will be
postponed until the Desserts.

This chapter is not intended to teach you all the details of the wide variety
of approaches for solving ODEs, but rather to show you how Maple can be used
as an auxiliary tool to implement some of the more common methods. However,
the series approach to solving an ODE will be postponed until Chapter 2.

As you are probably aware, the subject of solving differential equations (or-
dinary and partial) is huge. This is exemplified by Daniel Zwillinger’s Handbook
of Differential Equations [Zwi89] which is 700 pages long. This reference book
is highly recommended for quickly looking up every known method of solution.

1.1 Linear ODEs with Constant Coefficients

Consider an nth order LODE of the general structure (the spatial coordinate x
being replaced by t for time-dependent problems),

dny

dxn
+ an−1(x)

dn−1y

dxn−1 + · · · + a1(x)
dy

dx
+ a0(x) y = f(x). (1.1)

If f(x) = 0, the ODE is said to be homogeneous, otherwise it is nonhomogeneous.
The first ODEs that physics and engineering students usually encounter are
differential equations with constant coefficients (a0, etc., independent of x)
which can be solved in closed form in terms of “elementary” (trigonometric,
logarithmic, exponential) functions using a variety of standard methods. In this
section, I will illustrate how Maple can be used to implement these methods,
bypassing the tedious intermediate steps involved in a hand calculation.

1.1.1 Dazzling Dsolve Debuts

One should never make one’s debut with a scandal.
One should reserve that to give an interest to one’s old age.
Oscar Wilde, Anglo-Irish playwright, author, The Picture of Dorian Gray (1891)

Whether the coefficients are constant or not, the dsolve command is dazzling
in its ability to solve LODEs. Before we tackle some physical examples, this
mathematical recipe briefly looks at what is possible with dsolve.

> restart:

Let’s begin with the general nonhomogeneous first order LODE given in ode1 .
> ode1:=diff(y(x),x)+a(x)*y(x)=f(x);

ode1 := (
d

dx
y(x)) + a(x) y(x) = f (x)

A standard method for solving ode1 is to first find its integrating factor, IF .
Loading the necessary DEtools package, the intfactor command yields IF .

> with(DEtools): IF:=intfactor(ode1);

IF := e(
∫

a(x) dx)

The integral in IF is performed by specifying a(x), e.g., a(x) = a, a constant,
and applying the value command to IF.

> a(x):=a: IF:=value(IF);

IF := e(a x)

Multiplying ode1 by IF , the first integral I1 is obtained by applying firint.
> I1:=firint(IF*ode1);

I1 := e(a x) y(x) −
∫

e(a x) f (x) dx + C1 = 0

Since the ODE is first order, one arbitrary coefficient C1 appears. To evaluate
the integral in I1 , f(x) must be given. Suppose, e.g., that f(x)=e−x. Forming
value(I1), the first integral is explicitly evaluated.

> f(x):=exp(-x): I1:=value(I1);

I1 := e(a x) y(x) − e(a x−x)

a − 1
+ C1 = 0

The general solution to ode1 , labeled y1 , follows on solving I1 for y(x).
> y1:=solve(I1,y(x));

y1 := −−e(a x−x) + C1 a − C1
e(a x) (a − 1)

The above approach has mimicked the major steps of a hand calculation. The
same basic result can be obtained more quickly by applying dsolve to ode1 .

> dsolve(ode1,y(x));

y(x) = (
e(x (a−1))

a − 1
+ C1 ) e(−a x)

A common ODE notation is to use primes, each prime ( ′ ) standing for d/dx.
If desired, this notation can be introduced into the ODE output by first loading
the PDEtools library package and entering the following declare command.

> with(PDEtools): declare(y(x),prime=x);

y(x ) will now be displayed as y , derivatives with respect to x of functions
of one variable will now be displayed with ′

Then entering, say, the nonhomogeneous second-order linear ODE ode2 with
constant coefficients, produces an output with the prime notation.

> ode2:=diff(y(x),x,x)+a1*diff(y(x),x)+a0*y(x)=F(x);

ode2 := y ′′ + a1 y ′ + a0 y = F (x)

An ODE such as ode2 can be classified by applying the odeadvisor command.

> odeadvisor(ode2);

[[ 2nd order , linear , nonhomogeneous]]

An even more important diagnostic tool is the infolevel[dsolve] command,
which will give information on what methods are used in attempting to solve the
ODE when dsolve is applied, even if unsuccessful. An integer between 1 and
5 must be specified, with generally more detailed information being provided
as the number is increased. On applying the dsolve command to ode2 , the
method of attack is summarized in the following output and, in this case, the
general solution y(x) given with two arbitrary coefficients C1 and C2 .

> infolevel[dsolve]:=5: dsolve(ode2,y(x));

Methods for second order ODEs:
— Trying classification methods —
trying a quadrature
trying high order exact linear fully integrable
trying differential order: 2; linear nonhomogeneous with symmetry [0,1]
trying a double symmetry of the form [xi=0, eta=F(x)]
Try solving first the homogeneous part of the ODE

−> Tackling the linear ODE “as given”:
checking if the LODE has constant coefficients
<− constant coefficients successful

<− successful solving of the linear ODE ”as given”
−> Determining now a particular solution to the nonhomogeneous ODE

building a particular solution using variation of parameters
particular solution has integrals (!)
−> trying a d’Alembertian particular solution free of integrals
<− no simpler d’Alembertian solution was found

<− solving first the homogeneous part of the ODE successful

y = e((−a1
2 +

√
a12−4 a0

2 ) x) C2 + e((−a1
2 −

√
a12−4 a0

2 ) x) C1

+

(∫
e(−

(−a1+
√

a12−4 a0) x
2 ) F (x) dx e(x

√
a12−4 a0)

−
∫

F (x) e(
(a1+

√
a12−4 a0) x

2 ) dx

)
e(−

(a1+
√

a12−4 a0) x
2 ) /√a1 2 − 4 a0

The corresponding homogeneous ODE (set F (x)=0) was first solved and then
a particular solution obtained for F (x) �= 0 using the variation of parameters
method. The homogeneous solution is obtained by assuming that y ∼ eλ x. This
yields λ2 +a1 λ+a0=0, which has two roots λ1 and λ2. So the general solution
of the homogeneous ODE is of the form C1 eλ1 x + C2 eλ2 x, the specific forms
of λ1 and λ2 being easily identified from the above Maple output. The variation
of parameters method then assumes that y(x)= C1(x) eλ1 x + C2(x) eλ2 x for
the complete LODE, and solves for the functions C1(x) and C2(x). See, for
example, [Ste87] for the details.

The infolevel[dsolve] command can be turned off by setting it to 0. The
prime notation can also be turned off by entering the command OFF.

> infolevel[dsolve]:=0: OFF:

Boundary value and initial value problems are important in engineering and
physics. As a simple example of the former, suppose, e.g., that a1=2, a0=3,
and F (x)=x + sin(x) e−x. Then ode2 reduces to the form given in ode3 .

> a1:=2: a0:=3: F(x):=x+sin(x)*exp(-x): ode3:=ode2;

ode3 := (
d2

dx2 y(x)) + 2 (
d

dx
y(x)) + 3 y(x) = x + sin(x) e(−x)

Suppose that the boundary conditions (bcs) are that y(0) = y(1) = 0. The
boundary value problem is solved for y(x) by entering the equation name and
bcs in the dsolve command as a “Maple set” (items enclosed in braces, { }).
Unlike a list, a set does not preserve order or repetition.

> dsolve({ode3,y(0)=0,y(1)=0},y(x));

y(x) = −1
9

e(−x) sin(
√

2 x) (2 cos(
√

2) + e + 9 sin(1))
sin(

√
2)

+
2
9

e(−x) cos(
√

2 x)

+
1
9

e(−x) (3 x ex − 2 ex + 9 sin(x))

Now, an initial value problem is considered. First, let’s turn the declare
command back on and replace x with the time variable t in ode3 .

> ON: ode4:=subs(x=t,ode3);

ode4 := yt, t + 2 yt + 3 y(t) = t + sin(t) e(−t)

Because the independent variable is no longer x, the output has used the no-
tation yt and yt, t to indicate 1st and 2nd time derivatives. Now, suppose that

y(0) = 100 and yt(0) = 10. ode4 is now solved for y(t), subject to the initial
conditions. The differential operator D is used to enter the derivative condition.

> dsolve({ode4,y(0)=100,D(y)(0)=10},y(t));

y(t)=
490
9

e(−t) sin(
√

2 t)
√

2+
902
9

e(−t) cos(
√

2 t)+
1
9
e(−t) (3 t et − 2 et+9 sin(t))

To simplify y(t), the last output (indicated by %) must first be assigned. If
this is not done, entering y(t) would generate the output y(t), not the above
answer. Then y(t) is (somewhat) simplified and assigned the name y4 .

> assign(%): y4:=simplify(y(t));

y4 :=
1
9

e(−t) (490 sin(
√

2 t)
√

2 + 902 cos(
√

2 t) + 3 t et − 2 et + 9 sin(t))

ODE systems may also be solved with the dsolve command. Consider the
coupled first-order time-dependent LODEs given in ode5a and ode5b.

> ode5a:=diff(x(t),t)=-3*x(t)+4*z(t)+(sin(t))ˆ2*exp(-2*t);

ode5a := xt = −3 x (t) + 4 z (t) + sin(t)2 e(−2 t)

> ode5b:=diff(z(t),t)=2*z(t)-5*x(t)+(cos(t))ˆ3;

ode5b := zt = 2 z (t) − 5 x (t) + cos(t)3

The set of ODEs, ode5a and ode5b, is solved for the set of unknown functions,
x(t) and z(t), subject to the initial conditions x(0)=100 and z(0)=10.

> sol:=dsolve({ode5a,ode5b,x(0)=100,z(0)=10},{x(t),z(t)});
The solution sol , whose lengthy output has been suppressed here in the text,
is assigned, and exponential terms collected in x(t) and z(t), the results being
labeled X and Z, respectively.

> assign(sol): X:=collect(x(t),exp); Z:=collect(z(t),exp);

X:= (−57231
7480

sin(
√

55 t

2
)
√

55 +
203147
2040

cos(
√

55 t

2
)) e(− t

2 )

+ (
1
6

cos(2 t) − 1
8
) e(−2 t) +

39
170

cos(t) +
3

170
sin(t) +

3
34

sin(3 t) +
5
34

cos(3 t)

Z := (
15727
1632

cos(
√

55 t

2
) − 90973

5280
sin(

√
55 t

2
)
√

55) e(− t
2 ) − 3

68
sin(t)

+
3
17

cos(t) +
3
17

cos(3 t) − 3
68

sin(3 t) + (− 1
12

sin(2 t) +
1
6

cos(2 t) − 5
32

) e(−2 t)

If you are not already dazzled by the power of dsolve in obtaining the above
result try, e.g., some other nonhomogeneous terms in the ODEs. Unlike with
the hand calculation, a new answer can be rapidly and accurately generated.

To this point, we have had no control over the method of solution. As shall
be demonstrated in some of the following recipes, various self-evident options
can be included in dsolve, such as series, laplace, numeric, etc.

1.1.2 The Tale of the Turbulent Tail

Nonsense. Space is blue and birds fly through it.
Felix Bloch, 1952 Nobel laureate in physics, Heisenberg and the early days of
quantum mechanics, Physics Today, December 1976

As a post-doctoral fellow at the Chadwick Laboratory at the University of
Liverpool (England), I enjoyed my leisure time on dark, rainy, often foggy,
winter nights playing badminton with my fellow physicists. The Tale of the
Turbulent Tail is inspired by those games of badminton played so long ago.

In an article contained in a delightful reprint collection entitled The Physics
of Sports [PLA92], Peastrel, Lynch and Armenti reported on their experimental
investigation of the aerodynamics of a badminton shuttlecock or “bird” falling
vertically from rest. The relevant data is reproduced in Table 1.1, the distance
y in meters that the bird fell in t seconds being given.

Table 1.1: Data for the falling badminton bird.

t 0.347 0.470 0.519 0.582 0.650 0.674 0.717 0.766
y 0.61 1.00 1.22 1.52 1.83 2.00 2.13 2.44
t 0.823 0.870 1.031 1.193 1.354 1.501 1.726 1.873
y 2.74 3.00 4.00 5.00 6.00 7.00 8.50 9.50

From the last pair of entries in the table, Peastrel et al calculated the terminal
velocity (occuring when the downward gravitational force balances the upward
drag force due to air resistance) to be (9.50 − 8.50)/(1.873 − 1.726) = 6.8 m/s.
The gravitational acceleration was taken to be g = 9.8 m/s2.

The investigators’ goal was to determine which law of air resistance could
best account for the experimental data, Stokes’s law or Newton’s law. For the
Stokes model, the drag force on a body of mass m moving with velocity v is
given by FStokes = −a m v, while for the Newton model, FNewton = −bm |v| v.
The positive coefficients a and b can be related to the terminal velocity and g.

After loading the plots library package, I will begin this recipe by first
> restart: with(plots):

entering the experimental data of Table 1.1 so that the predictions of the two
models can be tested for goodness of fit. The data is entered as a “list of lists.”
The first entry in each list is the time, the second entry the distance.

> data:=[[0.347,0.61],[0.470,1.00],[0.519,1.22],[0.582,1.52],

[0.650,1.83],[0.674,2.00],[0.717,2.13],[0.766,2.44],
[0.823,2.74],[0.870,3.00],[1.031,4.00],[1.193,5.00],
[1.354,6.00],[1.501,7.00],[1.726,8.50],[1.873,9.50]]:

Taking the gravitational force on the bird to be mg and assuming that Stokes’s
law of air resistance prevails (a comment (prefixed by the sharp symbol #) to
this effect is added to the end of the command line), the acceleration of the
bird is given by ode1 . This is a first-order linear nonhomogeneous ODE.

> ode1:=diff(v(t),t)=g-a*v(t); #Stokes’s resistance law

ode1 :=
d

dt
v(t) = g − a v(t)

After a sufficiently long time, the gravitational and drag forces will balance and
the bird will reach its terminal velocity V1 = g/a. So the unknown constant a
can be eliminated by substituting a = g/V1 into ode1 .

> ode1:=subs(a=g/V1,ode1);

ode1 :=
d

dt
v(t) = g − g v(t)

V1
Since the bird falls from rest, the initial value problem is solved by entering
ode1 and the initial condition v(0)=0 in the dsolve command as a Maple set.

> sol1:=dsolve({ode1,v(0)=0},v(t));

sol1 := v(t) = V1 − e(−
g t
V1 ) V1

To compare with the experimental data, we need to calculate the distance that
the bird falls in T seconds. For the Stokes model, this distance y1 is obtained
by integrating the right-hand side (rhs) of sol1 from t = 0 to T .

> y1:=int(rhs(sol1),t=0..T);

y1 :=
V1 (−V1 + g T + V1 e(−

g T
V1 ))

g
Let’s now look at Newton’s model of air resistance. Since the motion is

in one direction (down) only, the absolute value sign can be dropped, so that
FNewton =−b m v2. The acceleration of the bird is now given by ode2 . This is
a first order nonlinear nonhomogeneous ODE.

> ode2:=diff(v(t),t)=g-b*v(t)ˆ2; #Newton’s resistance law

ode2 :=
d

dt
v(t) = g − b v(t)2

The terminal velocity is now given by V2 =
√

g/b. The unknown constant b is
removed from the equation by substituting b = g/V2 2 into ode2 .

> ode2:=subs(b=g/V2ˆ2,ode2);

ode2 :=
d

dt
v(t) = g − g v(t)2

V2 2

To see what methods are used in trying to solve ode2 , the infolevel[dsolve]
command is entered and set to 2. Applying the dsolve command to ode2 and
taking the rhs, an analytic solution v2 is generated for the velocity v(t).

> infolevel[dsolve]:=2: v2:=rhs(dsolve(ode2,v(t)));

Methods for first order ODEs:
— Trying classification methods —
trying a quadrature
trying 1st order linear
trying Bernoulli
trying separable

<− separable successful

v2 :=
2V2 e(

2 t g
V2 + 2 C1 g

V2 )

1 + e(
2 t g
V2 + 2 C1 g

V2 )
− V2

Maple has used the fact that ode2 is separable. A separable ([Ste87]) first-order
nonlinear ODE can be written in the form dy/dx = g(x) f(y). It is solved by
separating variables and integrating, i.e.,

∫
dy/f(y)=

∫
g(x) dx.

The arbitrary constant C1 is determined by evaluating v2 at t=0, setting
the result to 0, and solving for C1. On some executions, C1 involves I ≡ √−1,
so to ensure a real velocity v2 , the complex evaluation command (evalc breaks
a complex quantity into real and imaginary parts) is applied to v2 .

> _C1:=solve(eval(v2,t=0)=0,_C1); v2:=evalc(v2);

C1 := 0 v2 :=
2V2 e(

2 t g
V2 )

1 + e(
2 t g
V2 )

− V2

The distance y2 that the bird falls in T seconds, according to Newton’s model,
is now calculated by integrating v2 assuming that g > 0, V2 > 0, and T > 0.

> y2:=int(v2,t=0..T) assuming g>0,V2>0,T>0;

y2 :=
V2 (V2 ln(1 + e(

2 T g
V2 )) − g T − V2 ln(2))

g

So, which is the better model equation, y1 or y2? Entering the values of g, V1 ,
and V2 , the forms of y1 and y2 are then as follows:

> g:=9.8: V1:=6.8: V2:=6.8: y1:=y1; y2:=evalf(y2);

y1 := −4.718367347 + 6.800000000 T + 4.718367347 e(−1.441176471 T )

y2 := −3.270523024 + 4.718367347 ln(1. + e(2.882352940 T )) − 6.800000000 T
The experimental data is now plotted in gr1, the data points being represented
by size 14 (red by default) circles.

> gr1:=plot(data,style=point,symbol=circle,symbolsize=14):

The theoretical curves y1 and y2 are plotted in gr2 over the time interval T = 0
to 2 seconds. y1 is represented by a thick blue dashed line, y2 by a thick green
solid line. The entry linestyle=[3,1] would produce the same line styles.

> gr2:=plot([y1,y2],T=0..2,color=[blue,green],

linestyle=[DASH,SOLID],thickness=2):
The two graphs are then superimposed with the display command. The min-
imum number of tickmarks along the horizontal and vertical axes is specified,
and axis labels and a title added.

> display({gr1,gr2},tickmarks=[4,3],labels=["t","y"],title=
"Newton’s law (solid green), Stokes’s law (dashed blue)");

The resulting picture is shown in Figure 1.1. It is clear that Newton’s law of
resistance is the correct model for the falling badminton bird. Stokes’s law is

Newton’s law (solid line), Stokes’s law (dashed line)

0

5

0

y

0 5 1 1 5 2t

Figure 1.1: The experimental points agree with Newton’s resistance law.

known to apply to the laminar flow of air about a smooth moving object, while
Newton’s law applies to the turbulent flow around a non-smooth body. It is the
tail “feathers” of the moving badminton bird which create the turbulence.

1.1.3 This Bar Doesn’t Serve Drinks

By the time a bartender knows what drink a man will have
before he orders, there is little else about him worth knowing.
Don Marquis, American humorist and journalist, (1878–1937)

This recipe involves a bar undergoing rotational motion. No, I don’t mean
that the local pub appears to be spinning because one has overindulged in liq-
uid refreshments after a tough exam! It has to do with the possible deflection
modes ([Mor48], [Hil57], [SR66]) of an originally straight uniform solid bar of
length L which rotates in a horizontal plane about one end (say x = 0) with
uniform angular velocity ω. Below a critical frequency ω1, the bar remains
straight, but on exceeding ω1 it starts pulsating and its shape changes. On fur-
ther increasing ω, another critical value ω2 is reached, above which the shape
changes again, and so on. The values of the critical frequencies and the de-
flection modes depend on how the ends of the bar are supported. Here, I will
consider one possibility, leaving it to you to explore other boundary conditions.

For small deflections y(x) from the straight shape, the governing 4th order
LODE (each prime denoting an x derivative) is

y ′′′′ − k4y = 0, where k4 ≡ε ω2/(Y I). (1.2)

Here ε is the mass per unit length, Y is Young’s modulus, and I is the moment
of inertia of a cross-section of the shaft about an axis perpendicular to the
xy plane. To solve the boundary value problem, four boundary conditions are
required for the homogeneous fourth-order equation, two conditions for each

end of the bar. I will consider the case when the pivot point at x = 0 is a
clamped end (both y and the slope y ′ vanish at x=0), while x=L is a free end
(both y ′′ (∝ bending moment) and y ′′′ (∝ shearing force) vanish at x=L).

The recipe begins with the loading of the plots library package and the
entry of the infolevel[dsolve] command. The PDE tools library is also
accessed and the declare command used to generate the prime notation.

> restart: with(plots): infolevel[dsolve]:=2:

> with(PDEtools): declare(y(x),prime=x);

The governing ODE (1.2) is entered. Instead of inputting the fourth derivative
as diff(y(x),x,x,x,x), the short-cut entry diff(y(x),x$4) is used.

> ode:=diff(y(x),x$4)-kˆ4*y(x)=0;

ode := y ′′′′ − k4 y = 0
The general solution of ode is obtained using the dsolve command and the
rhs of the result taken. Maple recognizes that ode is a LODE with constant
coefficients and makes use of this fact to solve the equation.

> y:=rhs(dsolve(ode,y(x)));

Methods for high order ODEs:
— Trying classification methods —
trying a quadrature
checking if the LODE has constant coefficients
<− constant coefficients successful

y := C1 e(−k x) + C2 e(k x) + C3 sin(k x) + C4 cos(k x)
Clamped-end boundary conditions are applied at x = 0 in bc1 and bc2 , and
free-end boundary conditions at x=L in bc3 and bc4 .

> bc1:=eval(y,x=0)=0; bc2:=eval(diff(y,x),x=0)=0;

bc1 := C1 + C2 + C4 = 0
bc2 := − C1 k + C2 k + C3 k = 0

> bc3:=eval(diff(y,x$2),x=L)=0; bc4:=eval(diff(y,x$3),x=L)=0;

bc3 := C1 k2 e(−k L) + C2 k2 e(k L) − C3 sin(k L) k2 − C4 cos(k L) k2 = 0

bc4 := − C1 k3 e(−k L) + C2 k3 e(k L) − C3 cos(k L) k3 + C4 sin(k L) k3 = 0
Applying the analytic solve command, the set of four boundary conditions are
solved for the set of four unknown coefficients. All the coefficients are zero,
which corresponds to the trivial solution y(x)=0, i.e., the straight bar profile.

> solve({bc1,bc2,bc3,bc4},{_C1,_C2,_C3,_C4});
{ C1 = 0, C4 = 0, C3 = 0, C2 = 0}

To obtain a non-trivial result, let’s first solve the set of three boundary condi-
tions, bc1 , bc2 , and bc4 , for C1, C2, and C4 and assign the solution sol .

> sol:=solve({bc1,bc2,bc4},{_C1,_C2,_C4}); assign(sol):

C1, C2, and C4 are automatically substituted into the remaining boundary
condition, bc3 , which is divided by k2 and simplified.

> bc3:=simplify(bc3/kˆ2);

bc3 := −2 C3 (e(−k L) cos(k L) + 2 + e(k L) cos(k L))
e(−k L) − e(k L) + 2 sin(k L)

= 0

For a non-trivial solution, C3 �= 0, so the term containing cos(kL) must be
zero. The select command is used to extract this term from the lhs of bc3 .

> eq:=select(has,lhs(bc3),cos(k*L))=0;

eq := e(−k L) cos(k L) + 2 + e(k L) cos(k L) = 0
The expression eq is converted completely to trigonometric form and simplified.

> eq2:=simplify(convert(eq,trig)/2);

eq2 := cos(k L) cosh(k L) + 1 = 0
eq2 is a transcendental equation which must be solved numerically for k L. The
algebraic substitution kL=K is now made in eq2 .

> eq3:=algsubs(k*L=K,eq2);

eq3 := cos(K) cosh(K) + 1 = 0
The lhs of eq3 is plotted in Figure 1.2, the viewing range being controlled.

> plot(lhs(eq3),K=0..12,thickness=2,view=[0..12,-5..5]);

–4

0

4

6 12K

Figure 1.2: Graphically locating the zeros of the transcendental equation.

The approximate values of the zeros of eq3 can be determined by visual inspec-
tion. A slightly more accurate procedure is to place the cursor on the zero on
the computer screen and click the mouse. The horizontal and vertical coordi-
nates of the cursor location are displayed in a small viewing box at the top left
of the computer screen. Numerical values of the zeros to 10 digits accuracy can
be obtained by applying the numerical solving command,1 fsolve, to eq3 . If
lesser accuracy is desired, e.g., 5 digits, the Digits command can be used.

> Digits:=5:

A “functional”, or “arrow” operator2 f is introduced to systematically search
for the K zeros in the range 3 (n−1) to 3n using the fsolve command. Dividing
by L then yields the k zeros.

> f:=n->fsolve(eq3,K,3*(n-1)..3*n)/L:

When a number n is supplied, then subsequently entering f(n) will yield the
zero (if one exists) in the given range. For n=1 the search range is 0 to 3, for

1This command implements Newton’s method. [Ste87]
2Created on the keyboard with the hyphen (-) followed by the greater than symbol (>).

n=2 the range is 3 to 6, and so on. The sequence command, seq, is used in sol2
to generate the first four zeros of eq2 . Note that the k’s have been subscripted,
the input syntax being k[n]. The zeros are then assigned.

> sol2:=seq(k[n]=f(n),n=1..4); assign(sol2):

sol2 := k1 =
1.8751

L
, k2 =

4.6941
L

, k3 =
7.8548

L
, k4 =

10.996
L

The transcendental equation eq3 may be rewritten as cos(K) = −1/ cosh(K).
For large K, cosh(K) → ∞, so cos(K) → 0 and km → (m − 1/2) π/L for large
integer m. Setting a ≡√Y I/ε, the first four critical frequencies are calculated.

> critical_freq:=seq(omega[n]=a*k[n]ˆ2,n=1..4);

critical freq := ω1 =
3.5160 a

L2 , ω2 =
22.035 a

L2 , ω3 =
61.698 a

L2 , ω4 =
120.91 a

L2

The coefficient C3 is collected in y, the result evaluated at k = kn, and the
sequence command used to generate the profiles Yn of the first four deflection
modes. For brevity, only Y1 is shown here in the output of sol3 .

> sol3:=seq(Y[n]=eval(collect(y,_C3),k=k[n]),n=1..4);

sol3 := Y1 = (0.18111 e(
1.8751 x

L ) + 1.1811 e(− 1.8751 x
L ) + sin(

1.8751 x

L
)

− 1.3622 cos(
1.8751 x

L
)) C3 ,

sol3 is assigned and the first four possible deflection modes plotted for L = 5
and C3=1/5, the picture being shown in Figure 1.3. The solid curve with one
zero corresponds to n=1, the dashed curve with two zeros to n=2, etc.

> assign(sol3): L:=5: _C3:=1/5:

> plot([seq(Y[n],n=1..4)],x=0..5,color=[red,green,blue,cyan],

linestyle=[SOLID,DASH,DOT,SOLID],labels=["x","y"]);

Figure 1.3: First four deflection modes for the rotating bar.

1.1.4 Shake, Rattle, and Roll

Principles aren’t of much account anyway, except at election time.
After that you hang them up to let them season.
Mark Twain, American author, Municipal Corruption, speech, 4 Jan. 1901

It is election time on the planet Erehwon and the politicians are traveling far
and wide to woo potential voters. Presidential candidate Amai Koorc is trav-
eling on his own personal three-coach train which rides along a guide-way on
a cushion of air. The middle coach (mass M > m) is occupied by Koorc and
other political “heavies”, while the outer two coaches (each of mass m) contain
media “light weights”. Each outer coach is connected to the middle one by an
identical linear spring (i.e, governed by Hooke’s law) with spring constant K.

All three coaches were initially sitting stationary on a straight stretch of the
guide-way with the coupling springs unstretched, when fringe members of the
opposition party imparted an impulse to each coach causing the train to start
“rolling” down the track in an erratic fashion which shook up the politicians
and rattled the media. This recipe simulates the motion of the train.

To simplify the final results, it is assumed3 that K, m, and M are positive.
> restart: with(plots): assume(K>0,m>0,M>0):

Using Hooke’s law for the relative displacements x(t), y(t), and z(t) at time t
of the outer left, middle, and outer right coaches from equilibrium, Newton’s
second law yields the following three equations of motion.

> ode1:=m*diff(x(t),t,t)=K*(y(t)-x(t));

ode1 := m (
d2

dt2
x (t)) = K (y(t) − x (t))

> ode2:=M*diff(y(t),t,t)=K*(z(t)-y(t))-K*(y(t)-x(t));

ode2 := M (
d2

dt2
y(t)) = K (z (t) − y(t)) − K (y(t) − x (t))

> ode3:=m*diff(z(t),t,t)=-K*(z(t)-y(t));

ode3 := m (
d2

dt2
z (t)) = −K (z (t) − y(t))

This is a system of three coupled second-order LODEs with constant coefficients.
Given the general initial condition (dots indicating time derivatives) x(0)=A,
y(0)=B, z(0)=C, ẋ(0)=V 1, ẏ(0)=V 2, and ż(0)=V 3,

> ic:=(x(0)=A,y(0)=B,z(0)=C,D(x)(0)=V1,D(y)(0)=V2,D(z)(0)=V3):

it would be very tedious to solve for x(t), y(t), and z(t) by hand. Even Maple
needs some guidance here as to what method is best to use. If no method

3Note that assume applies the assumption throughout the entire worksheet, whereas
assuming applies it only in the command line that it is used. By default, assumed quan-
tities have attached “trailing tildes” in the output. The tildes can be removed by inserting
interface(showassumed=0) prior to the assumption. If desired, they can be removed from
all worksheets by clicking on File in the tool bar, then on Preferences, I/O Display, No
Annotation, and Apply Globally.

option is specified, applying the dsolve command to the three ODEs, sub-
ject to the initial condition, would yield a long “messy” result. Including the
method=laplace option, which makes use of the Laplace transform method,4

yields a somewhat simpler, but still lengthy, answer (not shown here).
> sol:=dsolve({ode1,ode2,ode3,ic},{x(t),y(t),z(t)},

method=laplace);
Using the list format [sin,cos], sine terms are now collected first in sol, and
then cosine terms. Since the analytic expressions are still lengthy, only x(t) is
displayed here in the text.

> sol2:=collect(sol,[sin,cos]);

sol2 := {x (t) =
1
2

√
m

K
(V1 − V3 ) sin(

√
K

m
t) + (

A

2
− C

2
) cos(

√
K

m
t)

+
1
2

(
M

2 m + M
)(3/2)

√
m

K
sin(

√
K (2 m + M)

M m
t) (V3 − 2V2 + V1 )

+
1
2

M (−2 B + C + A) cos(

√
K (2 m + M)

M m
t)

2 m + M

+
M (2 B + 2 tV2 ) + 2m (tV1 + C + A + tV3 )

2 (2 m + M)
, · · · · · ·}

Recalling that x(t) represents the displacement of the first outer coach from
equilibrium, it can be seen that the motion is built up of two parts. The trig
terms represent an oscillatory motion of the coach, while the last term corre-
sponds to a translational motion as time t increases. The oscillatory part is gov-
erned by two characteristic frequencies, ω=

√
K/m and

√
K (2 m + M)/(M m).

The solution sol2 is now assigned and a functional operator f created to
substitute parameter values into each solution and simplify the result.

> assign(sol2): f:=z->simplify(subs(par,z(t))):

The three coaches were initially in their equilibrium positions so A=B=C =0.
In some suitable set of Erehwonese units, the initial speeds imparted to the
coaches by the opposition gang were V 1 = 2/3, V 2 = −1/3, and V 3 = 1/6.
Each coach containing light-weight media had a mass m=1 while the political
heavies’ coach had a mass M =2. The spring constant K =4.

> par:=(A=0,B=0,C=0,V1=2/3,V2=-1/3,V3=1/6,m=1,M=2,K=4):

If the first outer coach was initially located at X =1, the middle one at Y =2,
and the other outer coach at Z =3, their positions at time t are as follows.

> X:=1+f(x); Y:=2+f(y); Z:=3+f(z);

4The Laplace transform of x(t) is defined as L(x(t)) ≡ X(s) =
∫∞
0

x(t)e−stdt. Integrating
by parts and assuming e−st x(t) → 0 as t → ∞, then L(ẋ) = s X(s) − x(0), and L(ẍ) =
s2X(s)−sx(0)− ẋ(0). To solve a LODE with constant coefficients, one can Laplace transform
the LODE, solve the resulting algebraic equation for X(s), and then perform the inverse
transform to obtain x(t). Laplace transforms are discussed in Chapter 6.

X := 1 +
3
32

√
2 sin(2

√
2 t) +

1
8

sin(2 t) +
t

24

Y := 2 − 3
32

√
2 sin(2

√
2 t) +

t

24

Z := 3 +
3
32

√
2 sin(2

√
2 t) − 1

8
sin(2 t) +

t

24
The positions of the three coaches are now plotted in Figure 1.4 over the interval
t=0 to 20, the linestyles 1, 2, 3 generating solid, dotted, and dashed curves.

> plot([X,Y,Z],t=0..20,color=[red,blue,green],linestyle=
[1,2,3],thickness=2,tickmarks=[3,3],labels=["t","X,Y,Z"]);

1

2

3

4

X,Y,Z

0 10 20t

Figure 1.4: Positions of the three train coaches as a function of time.

The motion of the three coaches (represented by size 20 red boxes) is now
animated and can be observed by executing the following command line.

> animate({[X,0],[Y,0],[Z,0]},t=0..100,frames=200,style=
point,symbol=box,symbolsize=20,view=[0..8,-0.1..0.1],
tickmarks=[3,2],labels=["x"," "]);

You can explore other possible motions by changing the parameter values and
initial conditions. The time range and view may have to be altered.

1.1.5 “Resonances”, A Recipe by I. M. Curious

Training is everything. The peach was once a bitter almond;
cauliflower is nothing but cabbage with a college education.
Mark Twain, Pudd’nhead Wilson, ch. 5, (1894)

Over the years I have been fortunate to have one or more students in each
mathematical physics class who stood out due to the quality of their work and
their innate curiosity about the subject matter. As a representative of this elite

group, let me introduce Ms. I. M. Curious who will discuss in her own words the
elegant and thorough computer algebra solution that she generated in solving
the following problem involving a familiar nonhomogeneous LODE.

A mechanical system (of unit mass) experiences a Stokes’s law drag force
FStokes = −2 α ẋ, a Hooke’s law restoring force FHooke = −ω2x, and a driving
force FDrive = f cos(Ω t + δ). Here α is a positive constant, ω the natural
frequency, f the force amplitude, Ω the driving frequency, and δ a phase angle.

(a) Obtain the solution of the resulting LODE for α < ω (underdamping),
subject to the initial condition x(0) = X , ẋ(0) = V . Discuss the method
used by Maple to solve the LODE and the structure of the solution.

(b) Derive an expression for the absolute value A of the steady-state ampli-
tude as a function of Ω. Determine the resonant frequency Ωr at which
A has its maximum value Amax and evaluate Amax.

(c) Taking X =−1, V =2, ω =1, α=1/5, Ω=3, f =5, and δ =0, determine
x(t) and v(t) = ẋ(t) and plot the results. Approximately how long does
it take for steady-state to prevail? Evaluate Ωr and plot the resonance
curves A vs. Ω for some different f values.

And here is I. M. to discuss her solution.
“After loading the plots package, infolevel[dsolve]:=5 is entered so as

to provide maximum information on Maple’s ODE solving method.
> restart: with(plots): infolevel[dsolve]:=5:

To save on later typing, I introduce the following alias command. When a,
w, W, d, Wr, and Am are subsequently entered, the symbols α, ω, Ω, δ, Ωr, and
Amax will be respectively produced in the output.

> alias(alpha=a,omega=w,Omega=W,delta=d,Omega[r]=Wr,A[max]=Am);

α, ω, Ω, δ, Ωr, Amax

The Stokes, Hooke’s law, and driving forces are entered,
> F[Stokes]:=-2*a*diff(x(t),t); F[Hooke]:=-wˆ2*x(t);

F[Drive]:=f*cos(W*t+d);

FStokes := −2 α (
d

dt
x (t)) FHooke := −ω2 x (t) FDrive := f cos(Ω t + δ)

which are automatically substituted into Newton’s second law in ode.
> ode:=diff(x(t),t,t)-F[Stokes]-F[Hooke]=F[Drive];

ode := (
d2

dt2
x (t)) + 2α (

d

dt
x (t)) + ω2 x (t) = f cos(Ω t + δ)

Assuming that α > 0, ω > 0, and α < ω, ode is solved for x(t) subject to the
initial conditions and the rhs of the result taken. The list of mathematical
methods tried are given in the output, along with the solution labeled x1 .

> x1:=rhs(dsolve({ode,x(0)=X,D(x)(0)=V},x(t)))
assuming(a>0,w>0,a<w);

Methods for second order ODEs:

— Trying classification methods —
trying a quadrature
trying high order exact linear fully integrable
trying differential order: 2; linear nonhomogeneous with symmetry [0,1]
trying a double symmetry of the form [xi=0, eta=F(x)]
Try solving first the homogeneous part of the ODE
−> Tackling the linear ODE “as given”:

checking if the LODE has constant coefficients
<− constant coefficients successful

<− successful solving of the linear ODE “as given”
−> Determining now a particular solution to the nonhomogeneous ODE

building a particular solution using variation of parameters
<− solving first the homogeneous part of the ODE successful

x1 := e(−α t) sin(
√

ω2−α2 t)(−2 V ω2 Ω2 + Ω4 α X + V Ω4 − Ω3 f sin(δ)
+ 4 V Ω2 α2 − Ω2 f α cos(δ) − α ω2 f cos(δ) − 2 Ω2 α ω2 X

+ Ω f sin(δ) ω2 + 4 Ω2 α3 X − 2 Ω f sin(δ) α2 + α ω4 X + V ω4)

/(
√

ω2 − α2 (ω4 − 2 ω2 Ω2 + 4 Ω2 α2 + Ω4))

+ e(−α t)cos(
√

ω2−α2 t)(f cos(δ) Ω2 − 2 f α Ω sin(δ) − f cos(δ) ω2+X ω4

− 2 X ω2 Ω2 + 4 X Ω2 α2 + X Ω4) /(ω4 − 2 ω2 Ω2 + 4 Ω2 α2 + Ω4)

+
((ω2 − Ω2) cos(Ω t + δ) + 2α Ω sin(Ω t + δ)) f

Ω4 + (−2 ω2 + 4 α2) Ω2 + ω4

Maple has successfully solved the ODE, a task I would not have wished to do
by hand. It has recognized that ode is a second order nonhomogeneous LODE
with the corresponding homogeneous equation having constant coefficients. It
has solved ode using the method of variation of parameters. The solution x1
is made up of two parts, the transient terms involving the exponentials which
vanish (since α > 0) as t → ∞, and the remaining steady-state contribution
which survives in this limit. Note that the steady-state part doesn’t depend on
the initial conditions. I will now extract the steady-state solution xss from x1
by removing the terms in x1 which have e−α t.

> xss:=remove(has,x1,exp(-a*t));

xss :=
((ω2 − Ω2) cos(Ω t + δ) + 2α Ω sin(Ω t + δ)) f

Ω4 + (−2 ω2 + 4 α2) Ω2 + ω4

The amplitude A of the steady-state solution can be obtained as follows. The
amplitude corresponds to maximum displacement of the oscillator at which time
the velocity is zero. I now solve for a time T at which this occurs.

> T:=solve(diff(xss,t)=0,t);

T :=
−δ + arctan(

2 α Ω
ω2 − Ω2 )

Ω
A is then obtained by evaluating xss at t=T and simplifying with the symbolic

option which picks a particular solution branch of the square root, the sign of
which can be either plus or minus. The magnitude of A then follows on taking
the absolute value of the result. The answer agrees with that found in Fowles
and Cassiday[FC99].

> A:=simplify(eval(xss,t=T),symbolic): Amag:=abs(A);

Amag :=
∣∣∣∣ f√

ω4 − 2 ω2 Ω2 + 4 Ω2 α2 + Ω4

∣∣∣∣
To find the resonance frequency, Amag is differentiated with respect to Ω, set
equal to zero, and the result solved for the frequency.

> sol:=solve(diff(Amag,W)=0,W);

sol := 0,
√

ω2 − 2 α2, −√
ω2 − 2 α2

The resonance frequency Ωr must correspond to the non-zero positive square
root solution. Therefore, the second answer is selected in sol . The maximum
amplitude Amax is then obtained by evaluating Amag at Ω = Ωr and again
simplifying with the symbolic option.

> Wr:=sol[2]; Am:=simplify(eval(Amag,W=Wr),symbolic);

Ωr :=
√

ω2 − 2 α2

Amax :=
1
2

∣∣∣∣ f

α
√

ω2 − α2

∣∣∣∣
To answer part (c) of the problem, the given parameter values are entered.

> X:=-1: V:=2: w:=1: a:=1/5: W:=3: f:=5: d:=0:

The displacement X1 and velocity V1 are calculated, the radical expressions
being simplified with the radsimp command.

> X1:=radsimp(x1); V1:=radsimp(diff(x1,t));

X1 :=
6737
9816

e(− t
5 ) sin(

2
√

6 t

5
)
√

6 − 159
409

e(− t
5 ) cos(

2
√

6 t

5
)

− 250
409

cos(3 t) +
75
818

sin(3 t)

V1 :=
179
9816

e(− t
5 ) sin(

2
√

6 t

5
)
√

6 +
1411
818

e(− t
5 ) cos(

2
√

6 t

5
)

+
750
409

sin(3 t) +
225
818

cos(3 t)

It was not specified how the results are to be plotted, so I will create a 3-
dimensional picture in gr1 of X1 vs. V1 vs. t by using the spacecurve com-
mand. The time range is taken to be t = 0 to 40, 1000 plotting points are
selected, and the plot is enclosed in a viewing box.

> gr1:=spacecurve({[t,X1,V1]},t=0..40,numpoints=1000,
thickness=2,axes=box,labels=["t","x","v"]):

So that the starting point in the 3-dimensional trajectory can be easily identi-

fied, the textplot3d command is used in gr2 to place the word start (colored
red) in the vicinity of the starting point.

> gr2:=textplot3d([[0,X-0.15,V,"start"]],color=red):

The graphs are superimposed with the display command. The resulting pic-
ture (Figure 1.5) is enclosed in a viewing box which may be rotated by clicking
on the computer plot and dragging with the mouse. In the upper left-hand cor-
ner of the computer screen, a small window indicates the angular coordinates
of the viewing box. Here the orientation of the box is set to be (45◦, 45◦).

> display({gr1,gr2},orientation=[45,45],tickmarks=[3,3,3]);

start

0

20

40

t

–1

0

1
x

–2

0

2

v

Figure 1.5: Space curve showing X1 vs. V1 vs. t.

The wild gyrations seen in Figure 1.5 for small t are associated with the tran-
sient, while steady-state corresponds to the uniform, cyclic behavior at larger
t. To obtain an estimate of the time beyond which steady-state prevails, the
orientation of the viewing box can be changed to (-90◦, 0◦) so as to view X1
vs. t. Steady-state prevails for t greater than about 25 time units.

The resonance frequency is then evaluated in W1 and the value A1 of the
absolute amplitude calculated at this frequency.

> W1:=evalf(Wr); A1:=evalf(Am);

W1 := 0.9591663048 A1 := 12.75775908
The resonance frequency of about 0.96 is slightly less than the natural frequency
ω=1. To locate the resonance frequency on the graph of the resonance curves,
I will plot a thick, vertical, dashed, blue line of height A1 at W1 in L.

> L:=plot([[W1,0],[W1,A1]],color=blue,thickness=2,linestyle=3):

To free the driving frequency and the force amplitude from their previously
assigned values, they are now unassigned.

> unassign(’W’,’f’):

A functional operator pl is created for plotting Amag vs. Ω for different f
values over the frequency range Ω = 0 to 2 ω.

> pl:=i->plot(eval(Amag,f=i),W=0..2*w,thickness=2):

Making use of the sequence command, resonance curves are generated for f =
1, 2, 3, 4, 5 and displayed along with L in Figure 1.6. As I expected, the resonance
curves agree with those drawn in standard texts.”

>display({L,seq(pl(i),i=1..5)},tickmarks=[2,3],labels=[W,"A"]);

0

5

10

A

1 2Omega

Figure 1.6: Resonance curves for f =1 (bottom curve) to f =5 (top).

1.1.6 Mr. Dirac’s Famous Function

The man with a new idea is a crank until the idea succeeds.
Mark Twain, Following the Equator, ch. 32, (1897).

Consider a very long uniform horizontal beam glued to an elastic foundation,
the foundation exerting a Hooke’s law restoring force. Under the influence of
a steady point force of magnitude F exerted at x = 0, the static displacement
ψ(x) of the beam is governed by the nonhomogeneous LODE [Hil57],

Y I
d4ψ

dx4 + K ψ = F δ(x) (1.3)
where Y is Young’s modulus, I the moment of inertia of the beam’s cross-
section about a horizontal axis, K the spring constant, and δ(x) the Dirac delta
function. The delta function (δ function) is defined by

δ(x) = 0, x �= 0; δ(x) = ∞, x=0;
∫ β

−α

δ(x) dx = 1, (α, β > 0). (1.4)

The δ function should be regarded as the limit of, e.g, a Gaussian function5

which is made progressively taller and narrower in such a way that the area

5For the Gaussian, δ(x)= lim A→∞(A/
√

π) e−A2 x2
. The choice of functions is not unique.

under the curve remains equal to 1. Changing the argument of the δ-function
from x to x − ζ shifts the location of the δ-function from x=0 to x=ζ.

It should be noted that the δ-function has the dimensions x−1. It has a
number of properties, the most important of which is the “selection” property.
If F (x) is a smooth function (not another δ-function), then∫ ζ+β

ζ−α

F (x) δ(x − ζ) dx = F (ζ), (α, β > 0). (1.5)

The Dirac delta function concept may be generalized to higher dimensions, e.g.,
δ(�r ) ≡ δ(x) δ(y) δ(z) is a 3-dimensional δ-function located at x=y=z=0.

A solution of a linear ODE which has a delta function appearing in the
nonhomogeneous term is called a Green function. The goal of the following
recipe is to determine the static displacement of the beam due to the point
force, i.e., the Green function, for Equation (1.3) and plot it.

In the ODE (1.3), the moment of inertia I appears. This is a protected
symbol in Maple, representing

√−1. Since I is a commonly used symbol in
physics, it is often desirable to unprotect it. The following interface command
is used to let j =

√−1, thus freeing up I.
> restart: interface(imaginaryunit=j):

The LODE (1.3) is inputted, the command Dirac(x) being used to enter δ(x).
> ode:=Y*I*diff(psi(x),x$4)+K*psi(x)=F*Dirac(x);

ode := Y I (
d4

dx4 ψ(x)) + K ψ(x) = F Dirac(x)

The ODE ode can be simplified by introducing the parameter α=(Y I/K)1/4

and letting ψ(x)=(F/K) G(x), where G(x) is defined to be the Green function.
The transformations are substituted into ode, producing ode2 ,

> ode2:=subs({Y=alphaˆ4*K/I,psi(x)=F*G(x)/K},ode);

ode2 := α4 K (
∂4

∂x4 (
F G(x)

K
)) + F G(x) = F Dirac(x)

which is then divided by F in ode3 , and the result expanded.
> ode3:=expand(ode2/F);

ode3 := α4 (
d4

dx4 G(x)) + G(x) = Dirac(x)

For x < 0 and x > 0, the rhs of ode3 is equal to zero. The general solution of
ode3 is obtained, assuming that x < 0.

> sol:=rhs(dsolve(ode3,G(x))) assuming x<0;

sol := C1 e(
(1/2−1/2 j)

√
2 x

α ) + C2 e(
(−1/2−1/2 j)

√
2 x

α )

+ C3 e(
(−1/2+1/2 j)

√
2 x

α ) + C4 e(
(1/2+1/2 j)

√
2 x

α )

Clearly, the general solution for x> 0 will be of a similar structure. Let’s look
at the region to the left of the origin, assigning this portion of G the name GL.
Since the beam is very long, little error is committed by taking the beam to be

of infinite length. Physically, the displacement of the beam from equilibrium,
and therefore GL, should go to zero as x → −∞. To satisfy this asymptotic
boundary condition, only the two exponentials which have (1/2 − 1/2 j) and
(1/2 + 1/2 j) in the exponents should be kept. The other two exponentials
involving (−1/2 − 1/2 j) and (−1/2 + 1/2 j) diverge as x→−∞. The following
command selects the two desired exponential terms to keep in GL.

> GL:=select(has,sol,{1/2-j/2,1/2+j/2});

GL := C1 e(
(1/2−1/2 j)

√
2 x

α ) + C4 e(
(1/2+1/2 j)

√
2 x

α )

Similarly, applying the condition G→0 as x→∞, the select command is used
to obtain the form GR of the Green function to the right of the origin.

> GR:=select(has,sol,{-1/2-j/2,-1/2+j/2});

GR := C2 e(
(−1/2−1/2 j)

√
2 x

α ) + C3 e(
(−1/2+1/2 j)

√
2 x

α )

The four constants must now be evaluated. Since the displacement is physically
continuous at the origin, the condition GR−GL=0 at x=0 is entered in eq0.

> eq[0]:=simplify(eval(GR-GL,x=0))=0;

eq0 := C2 + C3 − C1 − C4 = 0
An operator f is formed to simplify the nth derivative of the difference GR − GL
evaluated at x=0. Both n and the difference s must be given.

> f:=(n,s)->simplify(alphaˆn*(eval(diff(GR-GL,x$n),x=0)=s));

f := (n, s) → simplify(αn ((
∂n

∂xn
(GR − GL))

x = 0
= s))

For n=1 and 2, s=0, i.e., the first and second derivatives of G are continuous at
x=0. The third derivative (n=3) is discontinuous at x=0, however. Mentally
integrating ode3 from the left to the right of the δ function yields s = 1/α4.
Making use of these results and f , the remaining three conditions are entered.

> eq[1]:=sqrt(2)*f(1,0); eq[2]:=f(2,0);

eq[3]:=sqrt(2)*f(3,1/alphaˆ 4);

eq1 := − C2 − C2 j + C3 j − C3 + C1 j − C1 − C4 − C4 j = 0

eq2 := ( C2 − C3 + C1 − C4 ) j = 0

eq3 := − C2 j + C2 + C3 + C3 j + C1 + C1 j − C4 j + C4 =
√

2
α

The set of four equations are solved for the four constants,
> sol2:=solve({seq(eq[n],n=0..3)},{_C1,_C2,_C3,_C4});

sol2 :=

{
C1 =

(
1
8

− 1
8

j)
√

2

α
, C2 =

(
1
8

+
1
8

j)
√

2

α
, C4 =

(
1
8

+
1
8

j)
√

2

α
,

C3 =
(
1
8

− 1
8

j)
√

2

α

}

and the solution sol2 assigned. The constants are automatically substituted into
GL and GR, which are put into real forms by applying the evalc command.

> assign(sol2): GL:=evalc(GL); GR:=evalc(GR);

GL :=
1
4

√
2 e(

√
2 x

2 α ) cos(
√

2 x

2 α
)

α
− 1

4

√
2 e(

√
2 x

2 α ) sin(
√

2 x

2 α
)

α

GR :=
1
4

√
2 e(−

√
2 x

2 α ) cos(
√

2 x

2 α
)

α
+

1
4

√
2 e(−

√
2 x

2 α ) sin(
√

2 x

2 α
)

α

Noting that there is a symmetry in the two forms (replacing x with −x in
GL yields GR), the Green function can be put into a more compact form by
replacing x by its absolute value, |x|, in GR, then factoring, and relabeling the
result as G. The static displacement of the beam is then ψ(x)=(f/K) G(x).

> G:=factor(subs(x=abs(x),GR));

G :=
1
4

√
2 e(−1/2

√
2 |x|
α ) (cos(

1
2

√
2 |x|
α

) + sin(
1
2

√
2 |x|
α

))

α
The Green function G is now plotted for the representative value α=1.

> plot(eval(G,alpha=1),x=-10..10,thickness=2,

labels=["x","G"]);

0

0.35

G

–10 x 10

Figure 1.7: Green function G for the infinite beam.

The result is shown in Figure 1.7. As expected, G and therefore the beam
displacement is localized near the origin where the point force is applied.

1.2 Linear ODEs with Variable Coefficients

In this section, linear ODEs with variable coefficients are considered. Our at-
tention will be focused on the second-order Sturm–Liouville (S-L) equation,

d

dx

[
p(x)

dy

dx

]
− q(x) y + λw(x) y = 0, (1.6)

where λ is a real parameter while p(x), q(x), w(x) are real functions and w(x)
is taken to be non-negative over the range x = a to b of interest. The S-L
equation, and its nonhomogeneous counterpart, plays a very important role in
the mathematical analysis of many physical problems, particularly for boundary
value problems where y or its first derivative vanish at a and b. In this case the
y ’s are referred to as the eigenfunctions and the λ ’s as the eigenvalues. For
certain choices of p, q, w, and λ, the general S-L equation reduces to specific
“well-known” LODEs such as Bessel’s equation, Legendre’s equation, Hermite’s
equation, etc. The standard approach to solving these equations is to seek an
infinite power series solution. In some cases (e.g., for Bessel’s equation) the
solution remains as an infinite series, while in other cases the infinite series
have to be truncated (e.g., for Legendre’s and Hermite’s equations) to finite
polynomials to avoid divergence problems for large x. In either case, these
solutions to special forms of the S-L equation are referred to as special functions
(e.g., Bessel functions, Legendre functions (polynomials), etc.). The detailed
series derivation of these special functions will be postponed until Chapter 2. In
this section, we shall be content to see what Maple reveals about some of these
special functions and how physical problems involving these functions can be
easily dealt with. You will encounter many more members of the S-L “family”
of special functions as you progress through the book. For example, in the
Entrees you will see that special functions play an extremely important role in
the solutions of linear PDEs such as the wave and diffusion equations because
these equations can be “separated” into systems of S-L ODEs.

1.2.1 Introducing the Sturm–Liouville Family

My books are water; those of the great geniuses is wine.
Everybody drinks water.
Mark Twain, Mark Twains Notebooks and Journals, Notebook 26, (1887)

Jennifer, a young mathematician at the Metropolis Institute of Technology
(MIT) and a firm believer in using computer algebra to teach applied math-
ematics to engineers and physicists, has kindly consented to provide us with
some of the computer algebra recipes that she has developed. In the following
recipe, she introduces two of the more “famous” members of the S-L family,
namely the Bessel and Legendre ODEs.

Jennifer creates two functional operators, SL for generating a specific ODE
from the S-L equation (1.6) on subsequently specifying the forms of p, q, w

and the parameter L (short for λ), and Y for generating the general analytic
solution of a specified ode and taking the right-hand side of the result.

> restart:

> SL:=(p,q,w,L)->diff(p*diff(y(x),x),x)-q*y(x)+L*w*y(x)=0;

SL := (p, q, w, L) → (
d

dx
(p (

d

dx
y(x)))) − q y(x) + L w y(x) = 0

> Y:=ode->rhs(dsolve(ode,y(x))):

According to standard texts (e.g., [MW71], [Boa83]), Bessel’s’ equation should
result on choosing p=x, q =−x, w =−1/x, and L ≡ λ=−n2. Entering these
specific forms as arguments in SL produces ode1 .

> ode1:=SL(x,-x,1/x,-nˆ2);

ode1 := (
d

dx
y(x)) + x (

d2

dx2 y(x)) + x y(x) − n2 y(x)
x

= 0

Loading the DEtools library package, Jennifer enters the following odeadvisor
command line to see if Maple can classify ode1 and offer any information about
this equation. The ODE is identified as Bessel’s equation and inclusion of the
help option causes a help page to be opened with information on this equation.
She adds a comment to close the help page when finished reading it.

> with(DEtools): odeadvisor(ode1,help); #close Help page

[ Bessel ]
Entering the infolevel[dsolve] command so as to gain information about
the methods tried in attempting to solve the ODE, she then uses the arrow
operator Y to solve ode1 .

> infolevel[dsolve]:=2: y1:=Y(ode1);

Methods for second order ODEs:
— Trying classification methods —
trying a quadrature
checking if the LODE has constant coefficients
checking if the LODE is of Euler type
trying a symmetry of the form [xi=0, eta=F(x)]
checking if the LODE is missing ’y’
−> Trying a Liouvillian solution using Kovacic’s algorithm
<− No Liouvillian solutions exists
−> Trying a solution in terms of special functions:

−> Bessel
<− Bessel successful

<− special function solution successful
y1 := C1 BesselJ(n, x) + C2 BesselY(n, x)

After identifying the ODE as second order, Maple tries various approaches be-
fore seeking and obtaining a general special function solution in terms of Bessel
functions. Highlighting, say, BesselJ in the output with the mouse and clicking
on “Help on BesselJ”, opens a help page on Bessel functions. BesselJ(n, x) is

identified as a Bessel function of the first kind of order n, while BesselY(n, x)
is a Bessel function of the second kind. In standard mathematical notation,
they are usually written as Jn(x) and Yn(x), respectively. Before delving into
some of the properties of Bessel functions, Jennifer notes that on executing the
command ?inifcns, a help page is opened which lists all the functions known
to Maple and has hyperlinks to associated help pages (e.g., for BesselJ).

> ?inifcns;

Closing the help page, Jennifer forms an arrow operator f for plotting a se-
quence of four (the number can be easily changed) Bessel functions with integer
subscript. The name N (BesselJ or BesselY) of the Bessel function must be
supplied along with the horizontal range x=x1 to x2 over which the function
is to be plotted. The four curves are given different colors and line styles. Line
style 4 produces a DASH-DOT curve. The vertical range is limited with the
view command to −1 to +1 because the Yn(x) become infinitely large at x=0
if this point is included in the range of interest.

> f:=(N,x1,x2)->plot([seq(N(n,x),n=0..3)],x=x1..x2,thickness=2,
color=[red,blue,green,cyan],linestyle=[1,2,3,4],
view=[x1..x2,-1..1],tickmarks=[2,2]):

Making use of f, the Jn(x) and Yn(x) are plotted for x=x1=0 to x2=20.
> f(BesselJ,0,20); f(BesselY,0,20);

–1

0

1

10 20
x

–1

1

10 20
x

Figure 1.8: Left plot: Left to right, J0 to J3. Right: Left to right, Y0 to Y3.

The Bessel functions J0(x), J1(x), J2(x), and J3(x) are ordered from left to
right in the left plot of Figure 1.8. At x=0, J0 has the value 1, while all other
positive integer Jn are equal to zero there. The Bessel functions Y0(x), Y1(x),
Y2(x), and Y3(x), with the same ordering, are shown on the right of Figure 1.8.
The Yn(x) diverge to −∞ at x=0, so are rejected in physical problems where
the origin is part of the range of interest.

Of particular importance in solving problems involving Bessel functions is
knowing the locations of the zeros. The first 5 zeros of, e.g., J1 are now ob-
tained with the following BesselJZeros command. Note that the argument 1
is expressed in floating point form, i.e., 1.0, in order to numerically evaluate the
zeros. The command BesselYZeros will generate the zeros of the Yn(x).

> Zeros_J1:=BesselJZeros(1.0,1..5);

Zeros J1 := 3.831705970, 7.015586670, 10.17346814, 13.32369194, 16.47063005

For large x, both the Jn(x) and Yn(x) have the appearance of slowly decreasing
sine or cosine functions. Jennifer confirms this conjecture by Taylor expanding,
e.g., J1(x) about x=+∞ and keeping the first term in the expansion.

> taylor(BesselJ(1,x),x=infinity,1);

−
√

2 cos(x +
π

4
)
√

1
x√

π
+ O((

1
x

)(3/2))

The “order of” term, O((1/x)3/2), is removed with convert(%,polynom), the
“ditto operator”, %, applying the command to the previously executed result.6

> J1_asymptotic:=convert(%,polynom);

J1 asymptotic := −
√

2 cos(x +
π

4
)
√

1
x√

π
Asymptotically, J1(x) behaves like a cosine function whose amplitude decreases
like 1/

√
x. The Bessel functions have many important properties which are too

numerous for Jennifer to explore here. For an exhaustive list of the properties
of all special functions, she refers the reader to the voluminous Handbook of
Mathematical Functions [AS72]. One important property shared by special
functions is that they satisfy recurrence relations, relating functions of different
orders. Here’s an example of a recurrence relation for the Jn(x). This recurrence
relation relates Bessel functions of orders n−1, n+1, and n for arbitrary x.

> recurrence:=BesselJ(n-1,x)+BesselJ(n+1,x)

=simplify(BesselJ(n-1,x)+BesselJ(n+1,x));

recurrence := BesselJ(n − 1, x) + BesselJ(n + 1, x) =
2 n BesselJ(n, x)

x
Now, Jennifer introduces Legendre’s ODE by entering the Sturm–Liouville

arrow operator SL with p = 1−x2, q = 0, w = 1, and L ≡ λ = n (n + 1). The
differential equation ode2 is then solved and identified as Legendre’s equation.

> ode2:=SL(1-xˆ2,0,1,n*(n+1)); y2:=Y(ode2);

ode2 := −2 x (
d

dx
y(x)) + (1 − x2) (

d2

dx2 y(x)) + n (n + 1) y(x) = 0

Methods for second order ODEs:
...................................................................

6To recall the second previous result, use two percent signs, %%, and so on.

−> Trying a solution in terms of special functions:
−> Bessel
−> elliptic
−> Legendre
<− Legendre successful

<− special function solution successful
y2 := C1 LegendreP(n, x) + C2 LegendreQ(n, x)

Highlighting LegendreP with the mouse and opening up the associated help page
reveals that LegendreP(n, x) is the Legendre function of the first kind of order n,
while LegendreQ(n, x) is the Legendre function of the second kind. In standard
mathematical notation, they are written as Pn(x) and Qn(x), respectively. If
the odeadvisor command is applied to ode2 , the ODE is identified

> odeadvisor(ode2);

[ Gegenbauer ]
as a Gegenbauer equation, which has the general structure [AS72],

(1 − x2) y ′′ − (2 α + 1)x y ′ + n (n + 2 α) y = 0. (1.7)

Legendre’s ODE is a special case of Equation (1.7) with α=1/2.
The Legendre functions Pn(x) with n = 0 or a positive integer play an im-

portant role in boundary value problems involving spherical polar coordinates,
in which case x = cos(θ) where θ is the polar angle measured from the z-axis.
The polar angle runs from 0 to π radians, so x spans the range 1 to −1. For
n=0, 1, 2, ..., the P ’s reduce to the Legendre polynomials which Jennifer now
generates for n=0 to 5. The simplify command must be applied to produce
the finite polynomial forms.

> Ps:=seq(simplify(LegendreP(n,x)),n=0..5);

Ps := 1, x, −1
2

+
3 x2

2
,

5
2

x3 − 3
2

x,
3
8

+
35
8

x4 − 15
4

x2,
63
8

x5 − 35
4

x3 +
15
8

x

The polynomials may also be generated by loading the orthopoly library package
and using the syntax P(n,x) to enter the nth order Legendre polynomial.

> with(orthopoly): seq(P(n,x),n=0..5);

1, x, −1
2

+
3 x2

2
,

5
2

x3 − 3
2

x,
3
8

+
35
8

x4 − 15
4

x2,
63
8

x5 − 35
4

x3 +
15
8

x

To produce Qn(x) over the range x=−1 to 1, Jennifer enters EnvLegendreCut:
=1..infinity: which selects the desired mathematical branch7 of the function.
The Q ’s are then produced for n=0, 1, 2. They are expressible as combinations
of log functions.

> _EnvLegendreCut:=1..infinity:

> Qs:=seq(simplify(LegendreQ(n,x)),n=0..2);

7To plot Qn(x) for |x| > 1, enter EnvLegendreCut:=-1..1: and adjust the view.

Qs :=
1
2

ln(1 + x) − 1
2

ln(1 − x),
1
2

x ln(1 + x) − 1
2

x ln(1 − x) − 1,

−1
4

ln(1 + x) +
1
4

ln(1 − x) +
3
4

x2 ln(1 + x) − 3
4

x2 ln(1 − x) − 3 x

2
Using the arrow operator f, the Legendre functions of both kinds are now
plotted for n=0 to 3 over the horizontal range x=−1 to 1.

> f(LegendreP,-1,1); f(LegendreQ,-1,1);

–1

0

1

–1 1x

–1

0

1

–1 1x

Figure 1.9: Left plot: P0, P1, P2, P3. Right: Q0, Q1, Q2, Q3.

P0(x) (horizontal curve), P1(x) (diagonal curve), P2(x) (parabola), and P3(x)
(N-shape) are shown in the left plot of Figure 1.9. Q0(x) (solid curve), Q1(x)
(parabola), Q2(x) (N-shape), and Q3(x) (W-shape) are shown in the right plot
of the figure. The Pn are well behaved at x=±1, but the Qn diverge at these
points and must be rejected in physical problems which include x=±1 in the
range of interest.

An important general property that all solutions yn(x) of the Sturm–Liouville
equation corresponding to a given λn possess is orthogonality. Provided that
y(x), or y ′(x), or p(x) vanishes at the end points a and b of the range (referred
to as Sturm–Liouville boundary conditions), then∫ b

a

w(x) ym(x) yn(x) dx = 0, for m �= n. (1.8)

w(x) is referred to as the weight function. Noting that w(x)=1 for Legendre’s
equation and that p = 1 −x2 vanishes at x = a = −1 and at b = +1, the
orthogonality conditions

∫ 1
−1 Pm(x) Pn(x) dx = 0 and

∫ 1
−1 Qm(x) Qn(x) dx = 0

should prevail for m �= n. For specific values of m and n the former condition
is easy to prove, while proving the latter condition is more formidable.

Jennifer tackles the second orthogonality condition, taking m=3 and n=2.
She uses the “inert” form Int on the left of the following command line to

display the form of the integral and the “active” form on the right to evaluate
the integral. The answer is 0 as expected.

>orthog:=Int(LegendreQ(2,x)*LegendreQ(3,x),x=-1..1)

=int(simplify(LegendreQ(2,x)*LegendreQ(3,x)),x=-1..1);

orthog :=
∫ 1

−1
LegendreQ(2, x) LegendreQ(3, x) dx = 0

1.2.2 Onset of Bending in a Vertical Antenna

Truth is stranger than fiction, but it is because Fiction is obliged to
stick to possibilities; Truth isn’t.
Mark Twain, Following the Equator, ch. 15, (1897).

As a follow up to the last recipe, Jennifer has asked her class to examine the
stability of an antenna which consists of a thin vertical wire of length L and
uniform circular cross-section of radius a which is clamped at its lower end and
is free at its upper end. Let θ be the angular deflection of the antenna from
the vertical at a distance y from the top, Y the Young’s modulus, ρ the mass
density, and g the acceleration due to gravity. If the length L is small, the
antenna is “stable” in the vertical position, i.e., θ = 0 for all values of y. As
L increases, there is a critical value beyond which the antenna is unstable and
will bend from the vertical.

By considering the shear and gravitational forces on the wire, it can be
shown that the relevant LODE for small angular displacements θ is

d2θ

dy2 = −c2 y θ, where c =
2
a

√
ρ g

Y
. (1.9)

(a) Noting that θ = 0 at y = L (clamped at bottom end) and dθ/dy = 0 at
y = 0 (free at top), determine the solution of the ODE, identifying the
functions which occur. Express the solution in terms of Bessel functions.

(b) Prove that the critical length for bending is given by Lcr � (2.8/c)2/3.

(c) Determine Lcr for a steel (Y =2.1× 1011 N/m2 and ρ=7800 kg/m3) wire
of radius 1 mm. Take g=9.8 m/s2.

Jennifer has provided us with the following computer algebra solution taken
from her answer key. The governing ODE is entered.

> restart:

> ode:=diff(theta(y),y,y)=-cˆ2*y*theta(y);

ode :=
d2

dy2 θ(y) = −c2 y θ(y)

The ODE is solved for θ(y) subject to the boundary condition dθ/dy = 0 at
y=0 and the rhs of the solution taken.

> theta:=rhs(dsolve({ode,D(theta)(0)=0},theta(y)));

θ := C2
√

3 AiryAi(−(c2)(1/3) y) + C2 AiryBi(−(c2)(1/3) y)
Jennifer has not included the other boundary condition in the dsolve command
because then only the trivial solution θ = 0 would be produced. Highlighting
AiryAi in the output with the mouse, clicking on Help, and then on “Help on
AiryAi”, opens a help page which indicates that AiryAi is an Airy wave function.
AiryBi is another Airy function. According to Help, the Airy functions are
related to the Bessel functions Jn/3 where n is a positive or negative integer.

Using this information, θ is converted to Bessel functions of the first kind
and simplified assuming that y > 0 and c > 0. θ is expressed in terms of J(−1/3).

> theta:=simplify(convert(theta,BesselJ)) assuming y>0,c>0;

θ :=
2
3

C2
√

3 c(1/3) √
y BesselJ(

−1
3

,
2 c y(3/2)

3
)

The other boundary condition, θ(y=L) = 0, is now applied.
> bc:=eval(theta,y=L)=0;

bc :=
2
3

C2
√

3 c(1/3)
√

LBesselJ(
−1
3

,
2 c L(3/2)

3
) = 0

For θ to have a non-trivial solution, the coefficient C2 �= 0, so the boundary
condition must reduce to the Bessel function being equal to zero. The select
command is used to extract the Bessel function from the lhs of the bc and the
boundary condition is re-expressed as follows.

> bc:=select(has,lhs(bc),BesselJ)=0;

bc := BesselJ(
−1
3

,
2 c L(3/2)

3
) = 0

Clearly, the critical length for bending must correspond to finding the first zero
of the above Bessel function. The following op command is used to extract the
second operand from the lhs of bc.

> X:=op(2,lhs(bc));

X :=
2 c L(3/2)

3
The first zero of J(−1/3) is obtained.

> s:=BesselJZeros(evalf(-1/3),1);

s := 1.866350859
The critical value Lcr follows on setting X = s and solving for L. Only one of
the answers is real, the other two being complex.

> Lcr:=solve(X=s,L);

Lcr :=
1.986352708 (c2)(2/3)

c2 ,

(−0.7046901283 (c2)(1/3)

c
+

1.220559106 I (c2)(1/3)

c
)2,

(−0.7046901283 (c2)(1/3)

c
− 1.220559106 I (c2)(1/3)

c
)2

The first answer (the real one) is selected and simplified with the symbolic
option. This is the expression for the critical length.

> Lcr:=simplify(Lcr[1],symbolic);

Lcr :=
1.986352708

c(2/3)

To confirm that the critical length Lcr � (2.8/c)2/3, the numerator of Lcr is
extracted and raised to the 3/2 power in p. It is then evaluated to 2 digits
accuracy in p2, yielding 2.8 as desired.

> p:=(numer(Lcr))ˆ(3/2); p2:=evalf(p,2);

p := 2.799526291 p2 := 2.8
To determine Lcr for the steel wire, the given parameter values are entered.

> g:=9.8: a:=0.001: Y:=2.1*10ˆ11: rho:=7800:

The parameter c is evaluated and the critical length for bending of the steel
wire antenna determined.

> c:=evalf((2/a)*sqrt(rho*g/Y)): Lcr:=Lcr;

Lcr := 1.752545257
The antenna will not bend from the vertical if its length L < 1.75 meters.

1.2.3 The Quantum Oscillator

Anybody who is not shocked by this subject has failed to understand it.
Neils Bohr, 1922 Nobel laureate in physics, referring to quantum mechanics

Over the years I have had occasion to teach an introductory quantum mechanics
course which concentrates on how to do quantum mechanics, leaving it to our
departmental “guru” to deal with deeper quasi-philosophical questions of what
it means in a higher-level course. In the doing category, a standard problem
(see Griffiths [Gri95]) is to derive the probability distribution for the quantum
mechanical version of the one-dimensional simple harmonic oscillator. This may
be done either “algebraically” using “ladder operators” or by the more brute
force method of directly solving the time-independent Schrödinger equation
for the probability amplitude ψ. Once ψ is determined, the probability density
P (x) = |ψ(x)|2 may be calculated. The probability that a particle can be found
between, say x=a and x=b, is

∫ b

a
P (x) dx. Of course,

∫∞
−∞ P (x) dx = 1.

This recipe illustrates how P (x) may be painlessly derived, plotted, and
interpreted for the quantum oscillator starting with the Schrödinger equation.
The PDEtools package is loaded because it contains the dchange command
which will be used to change both the dependent and independent variables.

> restart: with(PDEtools):

The Schrödinger ODE is entered for a particle of mass m. Here E and V are the
total and potential energy and hbar ≡ h̄=h/(2π), where h is Planck’s constant.

> ode:=(hbarˆ2/(2*m))*diff(psi(x),x,x)+(E-V)*psi(x)=0;

ode :=
1
2

hbar2 (
d2

dx2 ψ(x))

m
+ (E − V ) ψ(x) = 0

For a particle experiencing a Hooke’s law restoring force, V = (1/2) m ω2 x2,
where ω is the frequency and x is the displacement of the particle from equi-
librium. On entering V , its form is automatically substituted into ode.

> V:=(1/2)*m*omegaˆ2*xˆ2: ode:=ode;

ode :=
1
2

hbar2 (
d2

dx2 ψ(x))

m
+ (E − m ω2 x2

2
) ψ(x) = 0

Since h̄ ω has the units of energy, let’s express E in these energy units, writing
E =(n + 1/2) h̄ ω. The form n + 1/2 of the “scale factor”, where the parameter
n remains to be determined, has been chosen for later convenience. At the same
time, ode is multiplied by (2/(h̄ω) and the result expanded in ode2 .

> E:=(n+1/2)*hbar*omega: ode2:=expand(2*ode/(hbar*omega));

ode2 :=
hbar (

d2

dx2 ψ(x))

ω m
+ 2 ψ(x) n + ψ(x) − ω ψ(x) m x2

hbar
= 0

The constants can be removed from ode2 by introducing a new independent
variable ζ defined by x =

√
h̄/(mω) ζ and also setting ψ(x) = f(ζ) e−ζ2/2.

This transformation of variables is now entered.
>tr:={x=sqrt(hbar/(m*omega))*zeta,psi(x)=f(zeta)*exp(-zetaˆ2/2)}:

Using the dchange command ode2 is expressed in terms of the new variables.
> ode3:=dchange(tr,ode2,[zeta,f(zeta)]);

ode3 := (
d2

dζ2 f (ζ)) e(−
ζ2

2 ) − 2 (
d

dζ
f (ζ)) ζ e(−

ζ2

2 ) + 2 f (ζ) e(−
ζ2

2 ) n = 0

Dividing ode3 by the common exponential factor, e−ζ2/2, yields ode4 ,
> ode4:=expand(ode3/exp(-zetaˆ2/2));

ode4 := (
d2

dζ2 f (ζ)) − 2 (
d

dζ
f (ζ)) ζ + 2 f (ζ) n = 0

which is the Hermite ODE. Hermite’s equation is another S-L ODE, being
obtained from (1.6) by setting p(ζ) = e−ζ2

, q(ζ) = 0, w(ζ) = e−ζ2
, and λ = 2n.

The general solution of ode4 is now sought.
> f:=rhs(dsolve(ode4,f(zeta)));

f :=
C1 e(

ζ2

2 ) WhittakerM(
n

2
+

1
4
,
1
4
, ζ2)

√
ζ

+
C2 e(

ζ2

2 ) WhittakerW(
n

2
+

1
4
,
1
4
, ζ2)

√
ζ

Surprisingly, the answer is not given in terms of Hermite functions, but in-
stead in terms of another special function, the Whittaker functions Mµ,ν(z)
and Wµ,ν(z), which satisfy Whittaker’s differential equation,

y ′′(z) + [−1/4 + µ/z + (1/4 − ν2)/z2] y(z) = 0, (1.10)

with µ=n/2+1/4, ν = 1/4, and z=ζ2 here. However, the WhittakerW function
can be converted to a Hermite function by using convert(f,Hermite).

> f:=convert(f,Hermite);

f :=
C1 e( ζ2

2 ) WhittakerM(
n

2
+

1
4
,

1
4
, ζ2)

√
ζ

+
C2 HermiteH(n,

√
ζ2) (ζ2)(1/4)

√
ζ 2n

To avoid divergence problems at ζ =±∞, the WhittakerM function is removed
> f2:=remove(has,f,WhittakerM);

f2 :=
C2 HermiteH(n,

√
ζ2) (ζ2)(1/4)

√
ζ 2n

from f , and the result simplified with the radsimp command.
> f2:=radsimp(f2);

f2 :=
C2 HermiteH(n, ζ)

2n

In terms of ζ, the probability amplitude is given by ψ = f2 e−ζ2/2.
> psi:=f2*exp(-zetaˆ2/2);

ψ :=
C2 HermiteH(n, ζ) e(− ζ2

2 )

2n

To be physically meaningful and to satisfy the normalization condition, ψ must
remain finite over the range ζ = −∞ to +∞. This can only be accomplished
if the WhittakerM function is removed (which has already been done) and n
takes on the values n=0, 1, 2, 3, . . .. In this case, the Hermite functions reduce
to the Hermite polynomials Hn(ζ). The Hermite polynomials can be readily
generated. Here are the first six.

> seq(H[n]=simplify(HermiteH(n,zeta)),n=0..5);

H0 = 1, H1 = 2 ζ, H2 = −2 + 4 ζ2, H3 = 8 ζ3 − 12 ζ,

H4 = 12 + 16 ζ4 − 48 ζ2, H5 = 32 ζ5 − 160 ζ3 + 120 ζ

The Hermite polynomials can also be generated by loading the orthopoly library
package and using the syntax H(n,x).

To achieve the normalization condition
∫∞

−∞ P (ζ) dζ = 1, the constant must
be chosen to be C2 =

√
2n/(

√
π n!). This form is substituted into ψ and the

probability density P calculated and simplified with respect to the exponentials.
> psi:=subs(_C2=sqrt(2ˆn/(sqrt(Pi)*n!)),psi);

> P:=simplify(psiˆ2,exp);

P :=
e(−ζ2) HermiteH(n, ζ)2

2n
√

π n!
To plot the probability density for a given value of n, P is turned into an
arrow operator with the unapply command.

> P:=unapply(P,n):

The probability distribution will now be explored for a specific n value, say
n=25. Let’s check that the right form of C2 was entered by integrating P (25)
over the entire infinite range of ζ. The answer should be 1, which it is.

> Prob:=int(P(25),zeta=-infinity..infinity);

Prob := 1
Classically, the energy conservation statement for the harmonic oscillator is

1
2
m v2 +

1
2
m ω2 x2 =

1
2
m v2 +

1
2
h̄ ω ζ2 = E = (n +

1
2
) h̄ ω,

where v is the speed. At the turning points v = 0, so that ζ = ±√
2 n + 1.

A functional operator L is now formed for calculating the magnitude of the
classical turning point for a specified n value.

> L:=n->sqrt(2*n+1):

Vertical lines are plotted at the two turning points along with P (25). A list
of lists is used for each of the vertical line entries. These lines are dashed and
colored blue, while the probability curve is a solid red curve.

> plot([[[L(25),0],[L(25),0.24]],[[-L(25),0],[-L(25),0.24]],

P(25)],zeta=-9..9,numpoints=2000,linestyle=[3,3,1],
color=[blue,blue,red],thickness=2,labels=["zeta","P"]);

zeta

0.1

0.2
P

–8 –6 –4 –2 2 4 6 8

Figure 1.10: Probability distribution for n=25.

The resulting picture is shown in Figure 1.10. From the figure, we can see that
the quantum oscillator has a non-zero probability of being outside the classical
turning points. This probability is now calculated for n = 25. The integral is
multiplied by 2, because there are two “tails” to the quantum distribution.

> Prob2:=evalf(2*int(P(25),zeta=L(25)..infinity));

Prob2 := 0.04506702177

The probability of being outside the classical range is about 0.045. A bizarre
feature that occurs for odd values of n is that the probability of finding the
particle at x = 0 (the center of the potential well) is 0. You are referred to
Griffiths for a more thorough discussion of the quantum oscillator. In that
reference, the probability distribution is drawn for n=100. It is a trivial task
to change n in the above recipe and replot the probability distribution. Try it!

1.2.4 Going Green, the Mathematician’s Way

Colorless green ideas sleep furiously.
Noam Chomsky, American linguistic scholar illustrating that grammatical struc-
ture is independent of meaning, Syntactic Structures, (1957)

Suppose that one desires to solve a nonhomogeneous Sturm–Liouville equation,

L[y(x)] ≡ d

dx

[
p(x)

dy

dx

]
− q(x) y = f(x). (1.11)

subject to S-L boundary conditions at x = a and b > a. The Green function
method [Zwi89] is to first solve the corresponding Green function ODE,

L[G(x|z)] = δ(x − z), with a < (x, z) < b, (1.12)

subject to the same boundary conditions, for the Green function G(x|z). Then
the solution to (1.11) is given by

y(x) =
∫ b

a

G(x|z) f(z) dz. (1.13)

The Green function approach, which is related ([SR66]) to the variation of pa-
rameters method, can be applied to other linear nonhomogeneous ODEs besides
the S-L equation. More importantly, it can be generalized to handle nonhomo-
geneous PDEs such as the nonhomogeneous wave and diffusion equations.

Here’s an introductory Green function method problem that I have often
assigned over the years in my mathematical physics course. The idea for it,
as well as for others, sprang from a table of Green functions that I stumbled
across while scanning through an old mathematical physics text by Margenau
and Murphy [MM57]. Some of these older books on mathematical techniques
for scientists are gold mines of useful information. Now to the problem.

Explicitly determine the Green function corresponding to the ODE,

x y ′′ + y ′ − n2/x = f(x), (1.14)
where n is a positive (non-zero) integer and the boundary conditions are that
y(0) must remain finite and y(1)=0. If n=1 and f(x) = −x2 for 0 ≤ x ≤ 1/2
and f(x) = −(1 − x)2 for 1/2 ≤ x ≤ 1, use this Green function to explicitly
determine y(x) and plot the result.

The solution is as follows. First let’s note that x y ′′ + y ′ = (x y ′) ′, so that
the ODE is a nonhomogeneous S-L equation with p(x)=x and q(x)=n2/x.

Now the ODE for the Green function is entered.
> restart:

> ode:=x*diff(G(x),x,x)+diff(G(x),x)-nˆ2*G(x)/x=Dirac(x-z);

ode := x (
d2

dx2 G(x)) + (
d

dx
G(x)) − n2 G(x)

x
= Dirac(x − z)

The general solution of the ODE is obtained, assuming that x < z. The general
solution for x > z is of a similar mathematical structure.

> sol:=rhs(dsolve(ode,G(x))) assuming x<z;

sol := C1 x(−n) + C2 xn

The Green function must satisfy the same boundary conditions as the solution
of the original equation. For x < z, we must have GL remain finite at x = 0.
Since n is a non-zero, positive integer, the term x−n must be removed from sol
to form GL.

> GL:=remove(has,sol,xˆ(-n));

GL := C2 xn

In some executions of the recipe, the coefficient C1 appears in the above out-
put instead of C2. To avoid difficulty with the solve command later, let’s
introduce our own coefficient C by using the operand command to extract the
second operand (xn) in GL and multiply it by C.

> GL:=C*op(2,GL);

GL := C xn

To form GR for x > z, let’s substitute C1 = A and C2 = B in sol .
> GR:=subs({_C1=A,_C2=B} sol);

GR := A x(−n) + B xn

To satisfy the bc at x=1, GR is evaluated at that point and set equal to zero.
> eq1:=eval(GR,x=1)=0;

eq1 := A + B = 0
At x=z, we have the continuity condition, GL=GR .

> eq2:=eval(GL=GR,x=z);

eq2 := C zn = A z(−n) + B zn

There is a discontinuity in slope at x=z, given by G ′
R(z)−G ′

L(z)=1/p(z)=1/z.
> eq3:=eval(diff((GR-GL),x),x=z)=1/z;

eq3 := −A z(−n) n

z
+

B zn n

z
− C zn n

z
=

1
z

The set of three equations (eq1 , eq2 , eq3 ) are solved for the set of three coeffi-
cients (A, B, C) and the solution sol2 is assigned.

> sol2:=solve({eq1,eq2,eq3},{A,B,C}); assign(sol2):

sol2 :=
{
B =

1
2 z(−n) n

, C =
−z(−n) + zn

2 zn z(−n) n
, A = − 1

2 z(−n) n

}
On simplifying GL and GR, the Green function is completely determined.

> GL:=simplify(GL); GR:=simplify(GR);

GL :=
xn (zn − z(−n))

2 n
GR :=

zn (xn − x(−n))
2 n

Using the piecewise command, the Green function can be written as a single
function G, which displays the general symmetry property G(x|z) = G(z|x).

> G:=piecewise(x<=z,GL,x>=z,GR);

G :=

⎧⎪⎨
⎪⎩

xn (zn − z(−n))
2 n

x ≤ z

zn (xn − x(−n))
2 n

z ≤ x

To continue with the solution, GL and GR are evaluated at n=1,

> GL1:=eval(GL,n=1); GR1:=eval(GR,n=1);

GL1 :=
x (z − 1

z
)

2
GR1 :=

z (x − 1
x

)

2
and the function f(z) entered as the two pieces f1 and f2 .

> f1:=-(zˆ2): f2:=-(1-z)ˆ2:

Now the integration in (1.13) has to be carried out. The integration is slightly
tricky because both G and f are piecewise functions. For x < 1/2, we have

y1 =
∫ x

0
GR f1 dz +

∫ 1/2

x

GL f1 dz +
∫ 1

1/2
GL f2 dz.

This integration is now carried out and the result simplified.

> y1:=simplify(int(f1*GR1,z=0..x)+int(f1*GL1,z=x..1/2)

+int(f2*GL1,z=1/2..1));

y1 :=
1
48

x (−6 x2 − 13 + 24 ln(2))

Similarly, the integration for x ≥ 1/2 is performed and simplified.

> y2:=simplify(int(f1*GR1,z=0..1/2)+int(f2*GR1,z=1/2..x)

+int(f2*GL1,z=x..1)) assuming x>0;

y2 := − 1
48

1 + 25 x2 + 6 x4 − 32 x3 + 24 x2 ln(x)
x

The complete solution y to the given nonhomogeneous ODE is pieced together,

> y:=piecewise(x<=1/2,y1,x>=1/2,y2);

y :=

⎧⎪⎨
⎪⎩

1
48

x (−6 x2 − 13 + 24 ln(2)) x ≤ 1
2

− 1
48

25 x2 + 1 + 6x4 − 32 x3 + 24 x2 ln(x)
x

1
2

≤ x

and plotted over the range x=0 to 1.

> plot(y,x=0..1,thickness=2,labels=["x","y"]);

The resulting profile is shown in Figure 1.11, thus finishing the problem.

Figure 1.11: Solution of the given nonhomogeneous S-L equation.

1.2.5 In Search of a More Stable Existence

If you pick up a starving dog and make him prosperous, he will not
bite you. This is the principal difference between a dog and a man.
Mark Twain, Pudd’nhead Wilson, ch. 16, (1894)

Not all LODEs, even if they are of the S-L type, are readily solved in terms
of either elementary or special functions. In this case, one can easily generate
a numerical solution using the numeric option of the dsolve command. To
illustrate this approach, consider the following mechanics example.

A particle of mass m is fastened to the origin by a spring obeying Hooke’s
law whose spring constant is k(t) = k0 sin(ω t). The mass is constrained to
move on a straight line through the origin. A critical frequency ωcr exists such
that for ω < ωcr the motion is unstable (displays unbounded growth as time
evolves), but is stable (bounded motion) for all frequencies above ωcr. Setting
r=ω

√
m/k0, determine the critical value rcr corresponding to ω=ωcr and plot

representative solutions for r just below and just above rcr. Take the initial
displacement to be x(0)=1 and the initial velocity to be ẋ(0)=0.1.

Applying Newton’s second law, the relevant LODE (dots indicating time
derivatives with respect to the time variable τ) is

m ẍ(τ) + k0 sin(ω τ) x(τ) = 0. (1.15)

Introducing the dimensionless time variable t = (
√

k0/m) τ , Equation (1.15)
can be rewritten (dots now indicating time derivatives with respect to t) as

ẍ(t) + sin(r t) x(t) = 0. (1.16)

After loading the plots package, a functional operator ode is formed for
generating Equation (1.16) for different input values of r.

> restart: with(plots):

> ode:=r->diff(x(t),t,t)+sin(r*t)*x(t)=0;

ode := r → (
d2

dt2
x (t)) + sin(r t) x (t) = 0

The initial condition is entered and an unsuccessful attempt (no output is gen-
erated) is made to analytically solve the ODE for arbitrary r.

> ic:=x(0)=1,D(x)(0)=0.1:

> dsolve({ode(r),ic},x(t));
An arrow operator sol is now created to find a numerical solution of the ODE
for a specified value of r. To achieve this, the option numeric8 is included
in the dsolve command line. The output=listprocedure option gives the
output as a list of equations of the form variable=procedure, where the left-
hand sides are the names of the independent variable, the dependent variable(s)
and derivatives, and the right-hand sides are procedures that can be used to
compute the corresponding solution components. This output form is most
useful when the returned procedure is to be used later as will be the case here.

> sol:=r->dsolve({ode(r),ic},x(t),numeric,output=listprocedure):
Using sol, a “do loop” is introduced for carrying out the repetitive numerical
calculation of x(t) at a specific time for systematically increasing values of r.
The general syntax for this common programming structure is:

for name from expression by expression to expression while expression do
statement sequence;
end do:

The statement sequence is the main body of the do loop giving the steps to be
followed in the calculation. In the following opening line of the do loop,

> for i from 1 to 200 do

name is the index i, the first expression is 1, the by expression is absent, the
third expression is 200, and there is no conditional while expression present.
Since the by expression is missing, the default is to increment i by 1 each
time the statement sequence is executed.

The statement sequence is now explained. Using sol, the following com-
mand line evaluates x(t) at arbitrary time t for r= i/100. Since i runs from
1 to 200, r will be incremented in steps of 0.01 from 0.01 to 2.00. The
concatenation operator || is used in assigning the ith result the name X||i.
This operator attaches the numbers 1 to 200 to X, so that the outputs are
assigned the names X1 ,X2 ,..., X200 . Concatenated names are useful for
employing in the sequence (seq) command.

> X||i:=eval(x(t),sol(i/100));

Then each X||i is evaluated at t=150, the result being labeled Y||i.
> Y||i:=X||i(150);

A large time has been chosen so that it will be easy to distinguish between the
unstable solutions which should have grown to large amplitudes and the stable

8The default numerical method is the Runge–Kutta–Fehlberg 45 ([BF89]) method (dis-
cussed in the Desserts). Other schemes are listed in the Maple Help under “dsolve,numeric”.

solutions which remain bounded. The do loop is now ended, the output being
suppressed with a line-ending colon.

> end do:

To plot the numerical output, 200 plotting points (pts) are formed using the
sequence command. The abscissa of the ith point is given by i/100, with i
running from 1 to 200. This generates the r values. Since the Y ||i for the
unstable solutions are extremely large, the log is taken of the absolute value of
the Y||i to form the ordinate values. For each plotting point, the abscissa and
ordinate values are put into a Maple list format.

> pts:=seq([i/100,log(abs(Y||i))],i=1..200):

Choosing a suitable view, the points are now plotted, the default being to join
consecutive points with a straight line. The result is shown in Figure 1.12.

> plot([pts],view=[0.01..2,-10..70],tickmarks=[4,3],

labels=["r",""]);

Figure 1.12: Log of the absolute value of x(t=150) versus r.

Remembering that the vertical scale is logarithmic, we can see that the region
of instability persists until r reaches slightly less than 1.5, then stability sets in
for larger r. That is to say rcr � 1.5. More precisely, looking at the numerical
output, rcr lies between 1.48 (unstable) and 1.49 (stable).

The odeplot command may be used to plot the numerical solution of an
ODE. An arrow operator gr is formed to apply this command to sol(r) for a
given r value. x(t) is plotted against the time t, the time range being from 0 to
150. To obtain an accurate curve, 2000 plotting points are chosen.

> gr:=r->odeplot(sol(r),[t,x(t)],0..150,numpoints=2000,

thickness=2,tickmarks=[3,3]):
Then entering gr(1.48) produces the figure shown on the left of Figure 1.13,
while gr(1.49) generates the figure on the right.

> gr(1.48); gr(1.49);

Figure 1.13: Left: Unstable solution for r=1.48. Right: Stable for r=1.49.

For r = 1.48 ever-growing oscillations occur, indicative of instability, while for
r=1.49 the oscillations remain bounded in amplitude, a sign of stability.

The motion of the particle along the x-axis for, e.g., r = 1.49 can now be
animated. The coordinates C (expressed as a list with the vertical coordinate
equal to 0) of the particle at arbitrary time t are evaluated using sol(1.49).
Then, e.g., the coordinates at time t=100, are obtained9 by entering C(100).

> C:=eval([x(t),0],sol(1.49)): C(100);

[1.52446967145678580, 0]
An operator gr2 is formed to plot a size 20 blue circle at coordinates C(i).

> gr2:=i->plot([C(i)],style=point,symbol=circle,

symbolsize=20,color=blue,labels=["x"," "]):
The command seq(gr2(i),i=0..250) plots graphs for t = i = 0, 1, 2, ...250.
Displaying this sequence with the option insequence=true generates a picture
which is animated by clicking on the computer plot and then on the start arrow.

> display(seq(gr2(i),i=0..250),insequence=true,

tickmarks=[3,3]);

1.3 Supplementary Recipes
01-S01: Newton’s Law of Cooling
Newton’s law of cooling is governed by the ODE, Ṫ =−k (T − Ts), where T is
the temperature of an object at time t, k is the positive cooling constant, and
Ts is the temperature of the object’s surroundings. If T (0) = T0, determine
T (t > 0). What information does including infolevel[dsolve]:=2: provide?

9Note that the numeric option of the dsolve command yields more digits in the output
than the “standard” default of 10 digits.

A horseshoe is originally 100 ◦C hotter than the surrounding air. After 15
minutes, the temperature difference has fallen to 60 ◦C. How long will it take
for the horseshoe to reach a temperature 10 ◦C above the surrounding air?

01-S02: Charging a Capacitor
A capacitor C is connected in series with a resistor R and a voltage source
V = V0(t/τ)2e−t/τ , where τ is a characteristic time.
(a) Making use of Ohm’s law for the voltage drop across R and Kirchoff’s

voltage sum rule, derive the ODE for the charge q(t) on C at time t.
Then, analytically solve the ODE for q(t), given that q(0)=0.

(b) Taking R = 5 ohms, C = 2 farads, V0 = 3 volts, and τ = 1 second, plot
q(t) for t = 0 to 20 seconds. At what time is q(t) a maximum and what
is the maximum charge? Answer this first qualitatively by clicking the
mouse cursor on the maximum, and then quantitatively.

01-S03: Radioactive Chain
An important radioactive chain involves the disintegration of the unstable 238U
nucleus. It decays via α-emission into 234Th, which in turn β-decays into 234Pa,
and so on until the stable isotope 206Pb is created. The decay rates of the
first three species (N is the number of atoms and λ the decay constant) in a
radioactive chain at time t can by described by

Ṅ1 = −λ1N1, Ṅ2 = λ1N1 − λ2N2, Ṅ3 = λ2N2 − λ3N3.

(a) If N1(0)=N , N2(0)=0, and N3(0)=0, determine N1(t) N2(t), and N3(t).

(b) The λ values for the uranium chain are vastly different, so for plotting
purposes consider a hypothetical radioactive chain for which N = 1000,
λ1 =1, λ2 =2, and λ3 =3. Plot N1(t) N2(t), and N3(t) in the same figure
for t = 0..5 using different colors and line styles for each curve. At what
time is N3 a maxima? What is N3 at this time?

01-S04: Stokes and Newton Join Forces
The drag force ([FC99]) on a sphere of diameter d moving with speed v is, in
general, given in SI units by Fdrag = −a v − b v2, with a = 1.55 × 10−4d and
b=0.22 d2. I.e., it is a combination of Stokes’s and Newton’s resistance laws
(a) A spherical mass m is dropped from rest. Derive and solve the relevant

nonlinear ODE for v(t). What method does Maple use to solve the ODE?

(b) A basketball (d=0.25, m=0.60), a raindrop (d=10−4, m=0.52 × 10−9),
and a soap bubble (d=10−2, m=10−7 ), are all dropped from rest. Taking
the gravitational acceleration g=9.8 m/s2, determine v(t) for each body.

(c) Plot each v(t), showing the approach to the terminal velocity V . Deter-
mine V for each body and the time it takes to come within 1% of V ?

01-S05: Exploring the RLC Series Circuit
(a) Derive the ODE for a series circuit consisting of a resistor R, an inductor

L, and a capacitor C and cast it into the form q̈ + 2 α q̇ + ω2 q=0, where
q(t) is the charge on C at time t, α=R/(2 L) and ω=1/

√
L C.

(b) Given q(0) = A, q̇(0) = B, solve the ODE for overdamping (α > ω),
underdamping (α < ω), and critical damping (α = ω). What fact does
Maple use in successfully arriving at each solution? Calculate the currents.

(c) Taking A=−1, B =2, ω =1, α=α1=1/5, and α=α2=2, evaluate q(t)
and the currents for the three cases. Plot the q(t) together using different
colors and linestyles for each case. Do the same for the currents.

01-S06: The Whirling Bar Revisited
Suppose that the rotating bar of recipe 01-1-3 has hinged-end (y and y ′′ (the
bending moment) are both zero) boundary conditions at each end. Obtain the
first four critical frequencies and corresponding shapes and plot the latter in a
single figure. Compare the lowest critical frequency to that in 01-1-3.

01-S07: Driven Couple Oscillators
Masses m1 and m2, with equilibrium positions at x = 1 and x = 5, are free to
move horizontally on a smooth surface (the x-axis). m1 is connected to a fixed
wall on its left by a linear spring (spring constant k) and on its right to m2
with an identical spring. A driving force F =f sin(ω t) acts to the right on m2.
Air resistance is present, given by Stokes’s drag law, Fdrag =−a v.

(a) Derive the governing ODEs for the displacements x1(t) and x2(t) of m1
and m2 from equilibrium. Taking m1 =2, m2 =1, k=1, a=1/100, ω= 2,
f = 2, x1(0) = 1/10, x2(0) = 0, ẋ1(0) = 0, and ẋ2(0) = 0, solve the ODEs
using the Laplace transform option. Express x1(t) and x2(t) in real forms.

(b) Extract the steady-state parts of x1 and x2 and plot them. Discuss the
results. Plot the transient expressions over a time interval for which the
transients become small. This will reveal the envelope of the transients.

(c) Animate the motion of m1 and m2 about their equilibrium positions over
a time interval for which the transients become small.

01-S08: Some Properties of the Delta Function
(a) Two common representations of the δ function are G(x)=(α/

√
π) e−α2 x2

and F (x) = sin(α x)/(π x) with α → ∞. Confirm that these representa-
tions are reasonable by (i) plotting G(x) and F (x) for increasing α,
(ii) showing that

∫∞
−∞ G(x) dx=1 and

∫∞
−∞ F (x) dx=1 for α > 0.

(b) Using Dirac(x-a), confirm the following: (i)
∫∞

−∞ δ(x − a) dx=1,
(ii)
∫∞

−∞ f(x) δ(x − a) dx=f(a), (iii)
∫∞

−∞ f(x) δ ′(x − a) dx=−f ′(a),
(iv)

∫∞
−∞ δ(b(x − a)) dx=1/|b|.

(c) Using the command Heaviside(x), create a rectangle h(x) of unit height
and with edges at x = ±1. Plot h(x) for x = −3 to 3, using constrained
scaling. Differentiate h(x) with respect to x and interpret the result.

01-S09: A Green Function
Derive the Green function G which is the solution to G ′′ + k2 G = −δ(x − ζ),
subject to the boundary conditions G(−1)=G(1) and G ′(−1)=G ′(1). Simplify

G as much as possible and write it as a piecewise function. Check that G obeys
the general symmetry property G(x|ζ)=G(ζ|x). At what k values does G have
singularities where G diverges?

01-S10: A Potpourri of General Solutions
Classify the following LODEs as to order, homogeneneous or nonhomogeneous,
and constant or variable coefficients. Obtain the general solution for each and
identify which part of the solution is the particular solution and which part is the
solution to the corresponding homogeneous ODE. Identify any non-elementary
functions in the solution. What method of attack does Maple use?

(a) x y ′ + (1 + x) y = ex;

(b) y ′′′ + x y = 0;

(c) y ′′ + 3 y ′ + 2 y = ex;

(d) x2 y ′′ − 6 y = x3 lnx;

(e) y ′′ − (3/x) y ′ + (15/(4 x2) +
√

x) y = 0;

(f) x2 y ′′ + (1 − 2 α) x y ′ + (k2 β2 x2 β + (α2 − p2 β2)) y = 0.

01-S11: Chebyshev Polynomials
Taking p =

√
1 − x2, q = 0, w = 1/

√
1 − x2, and λ = n2, show that the S-L

ODE yields the Chebyshev equation, (1 − x2) y ′′ − x y ′ + n2 y = 0. For n a
non-negative integer, Chebyshev’s ODE is known to have polynomial solutions
(the Chebyshev polynomials Tn(x)) defined over the range x=−1 to +1. Using
the command expand(ChebyshevT(n,x)), generate the Tn(x) for n=0 to 10.
Then use the dsolve command to obtain the general solution of the Chebyshev
ODE and show that it can be expressed in terms of the Tn(x). Demonstrate
that, e.g.,

∫ 1
−1 T9(x) T10(x) w(x) dx=0, i.e., the Chebyshev polynomials form a

set of orthogonal functions. Plot the first five polynomials for x=−1 to 1.

01-S12: The Growing Pendulum
Consider a pendulum which consists of a point mass m at the bottom end of
a light supporting rod of length L which is allowed to move in a vertical plane
about a pivot point at its top end. Suppose that L increases at a steady rate,
i.e., L = L0 + v t, where v > 0 is a constant speed and t the time.

(a) Letting the rod make an angle θ(t) with the vertical and neglecting drag,
use Newton’s second law for angular motion (d(I ω)/dt = T , where I is
the moment of inertia, ω the angular velocity, and T the torque) to derive
the LODE for small θ. Solve the ODE for θ(t), given θ(0)=Θ, θ̇(0)=0.
Simplify θ(t) and identify the functions which occur.

(b) Taking L0 =1 m, g=10 m/s2, Θ=π/6 rads, and v=0.5 m/s, plot θ(t) for
t=0 to 100 seconds. Describe the behavior of the period. Then animate
the motion of the pendulum arm (representing it as a thick line) over this
time interval taking 100 frames and using constrained scaling.

01-S13: Another Green Function
Assuming that m �= 0, derive the Green function G which is the solution to
(1 − x2) G ′′ − 2 x G ′ − m2 G/(1 − x2) = δ(x − z), subject to the boundary
conditions that G remains finite at the end points x=±1 of the range. Simplify
G as much as possible and write it as a piecewise function. Plot G over the
range x=−1 to 1, taking m=3 and z = 1/2.

01-S14: Going Green, Once Again
A light elastic string, under tension T and fixed at x=±L, is horizontal in the
absence of any applied forces. The string is embedded in an elastic membrane
which exerts a Hooke’s law (spring constant k) restoring force if the string is
displaced from equilibrium. If a force f e−β x2

is applied to the string, derive
the ODE for its shape y(x), assuming that y(x) is small. Using the Green
function method, determine y(x). The answer will involve the error function,
erf(z)≡ (2/

√
π
∫ x

0 e−t2 dt. Check your result by solving the original ODE with
Maple’s dsolve command, subject to the given boundary condition. Taking
T =100 N, k=10 N/m, L=1 m, β=1/100 m−2, and f =50 N, plot y(x).



Chapter 2

Applications of Series
In this chapter, it is assumed that you are already familiar with the basic
concepts of infinite series covered in standard calculus texts such as Calculus
by James Stewart [Ste87]. The emphasis here is on applications of series, the
topics being Taylor series, series solutions of LODEs, Fourier series, Legendre
and Bessel series, and summing series. Laurent series are covered in Chapter 5.

2.1 Taylor Series

The Taylor expansion of a function f(x)=f(x0 + h) about x0 is given by

f(x)=f(x0) + h f ′(x0) +
1
2!

h2 f ′′(x0) +
1
3!

h3 f ′′′(x0) + · · · . (2.1)

Thus, e.g., taking x0 =0, so h=x − x0 =x,

sin(x)=x − x3

3!
+

x5

5!
− x7

7!
+ · · ·=

∞∑
n=0

(−1)n x2n+1

(2n + 1)!
≡

∞∑
n=0

un.

Applying the ratio test, r ≡ |un+1/un|=x2/((2n + 3)(2n + 2)) → 0 as n → ∞.
Since r < 1, the series converges for all x, i.e., the radius of convergence is ∞.

The extension to functions of more than one variable is straightforward, e.g.,
for a function f(x, y)=f(x0 + h, y0 + k), the Taylor series is of the form,

f(x, y)=f(x0, y0)+
(

h
∂

∂x
+k

∂

∂y

)
f(x0, y0)+

1
2!

(
h

∂

∂x
+k

∂

∂y

)2

f(x0, y0)+· · · .

2.1.1 Polynomial Approximations

Every man sees in his relatives, and especially in his cousins,
a series of grotesque caricatures of himself.
H. L. Mencken, American journalist, (1880–1956)

In this recipe, I will demonstrate how the Taylor series expansion may be used
to obtain the smallest polynomial approximation to the integral

f =
∫ x

0
u4 eu2

du, valid within ±0.000001 over the range 0 ≤ x ≤ 1.

The desired accuracy A is entered, and then the integral f is displayed and
evaluated in the following command line.

> restart: A:=0.000001:

> f:=Int(uˆ4*exp(uˆ2),u=0..x)=int(uˆ4*exp(uˆ2),u=0..x);

f :=
∫ x

0
u4 e(u2) du =

1
2

x3 e(x2) − 3
4

x e(x2) − 3
8

I
√

π erf(x I)

The answer is given in terms of the error function, erf(z)=(2/
√

π)
∫ z

0 e−u2
du,

with z=I x=
√−1 x. The result can be expressed in terms of real quantities by

converting the rhs to an imaginary error function, erfi(x)=(2/
√

π)
∫ x

0 eu2
du.

> f:=convert(rhs(f),erfi);

f :=
1
2

x3 e(x2) − 3
4

x e(x2) +
3
8

√
π erfi(x)

Then f is simplified by collecting exponential terms.
> f:=collect(f,exp);

f := (
1
2

x3 − 3
4

x) e(x2) +
3
8

√
π erfi(x)

Although f can be evaluated numerically, a polynomial approximation to f is
now obtained. An operator F is formed which Taylor expands f in powers of
x about x=0, dropping terms of order xN+1.

> F:=N->taylor(f,x=0,N+1):

A second arrow operator F2 is created to remove the order of term in the series.
> F2:=N->convert(F(N),polynom):

Using F2(2*n+1), a sequence of polynomial approximations to the integral is
generated in T for n=0 to 10 (not all polynomials are shown here). Note that
the Taylor series expansion of the integral generates only odd order polynomials,
hence the argument 2 n + 1. For plotting purposes, T is assigned.

> T:=seq(t||(2*n+1)=F2(2*n+1),n=0..10); assign(T):

T := t1 = 0, t3 = 0, t5 =
x5

5
, t7 =

1
5

x5+
1
7

x7, t9 =
1
5

x5+
1
7

x7+
1
18

x9,

t11 =
1
5

x5+
1
7

x7+
1
18

x9+
1
66

x11, ............................................................,

t21 =
1
5

x5+
1
7

x7+
1
18

x9+
1
66

x11+
1

312
x13+

1
1800

x15+
1

12240
x17+

1
95760

x19

+
1

846720
x21

The difference between the exact integral f and each polynomial for n = 2
(yielding t5 ) and 10 (yielding t21 ) is plotted over the range x=0 to 1.

> plot([seq(f-t||(2*n+1),n=2..10)],x=0..1,view=[0..1,0..A],

axes=box,thickness=2);
Since the differences turn out to be positive, the vertical view has been taken to
be between 0 and A, the upper limit setting the desired accuracy. The sequence

of differences is shown in Figure 2.1, the difference corresponding to n = 2 on
the far left and that corresponding to n = 10 on the far right. Since the first
difference curve that remains completely within the vertical viewing range is
the curve corresponding to n=10, the smallest polynomial which is within the
desired accuracy over the range x=0 to 1 is given by t21 .

0

2e–07

4e–07

6e–07

8e–07

1e–06

0.2 0.4 0.6 0.8 1x

Figure 2.1: Differences between exact integral and polynomial approximations.

2.1.2 Finite Difference Approximations

I’m not a teacher: only a fellow-traveler of whom you asked the way.
I pointed ahead–ahead of myself as well as you.
George Bernard Shaw, Anglo-Irish playwright (1856–1950)

Replacing ordinary and partial derivatives by their finite difference approxi-
mations (FDAs) is useful in numerically solving ODEs and PDEs as will be
illustrated in Chapter 9. These approximations are based on a Taylor series
approach. For example, an FDA to y ′′(x) can be obtained by Taylor expanding
y(x ± h) about x for small h, viz.,

y(x ± h) = y(x) ± hy ′(x) +
1
2!

h2y ′′(x) ± 1
3!

h3y ′′′(x) + O(h4), (2.2)

where O(h4) indicates that terms of order h4 (and higher) are being neglected.
Forming the sum y(x + h) + y(x − h), and solving for y ′′, yields the FDA,

y ′′(x) = [y(x + h) + y(x − h) − 2y(x)]/h2 + O(h2). (2.3)
Eq. (2.3), known as the central difference approximation to the second deriva-
tive, was easily derived by hand. However, a computer algebra approach can
sometimes prove useful in deriving other FDAs as our mathematical physics
student, I. M. Curious, will now illustrate in answering the following problem.

(a) Show that to O(h4) an alternate FDA to the second derivative is given by

y ′′(x)=[−y(x+2h)+16y(x+h)−30y(x)+16y(x−h)−y(x−2h)]/(12h2).

Taking 12 digits accuracy, plot the difference between the FDA and the
exact y ′′ for y=x5.1 over the range x=0.1 to 10 for h=0.1 and h=0.05.

(b) Show that an FDA for fxxyy ≡ ∂ 4f(x, y)/∂x 2∂y 2 is given by

fxxyy =[f(x+h, y + k)+f(x − h, y+k)+f(x+h, y − k)+f(x − h, y − k)
−2 (f(x+h, y)+f(x − h, y)+f(x, y+k)+f(x, y − k))+4 f(x, y)]/(h2k2),

and produce a 3-dimensional plot of the difference between the FDA and
the exact fourth derivative for f = sin(x − y) e−x2 y2

. Take h = k = 0.05,
12 digits accuracy, and a plotting range x=−3...3, y=−3...3.

I. M. begins her recipe for part (a) by setting the accuracy to 12 digits.
> restart: Digits:=12:

Noting that the wording of the problem implies that terms of order h6 have
been dropped in the Taylor series, she creates an arrow operator t to Taylor
expand y(x+ a h) in powers of h about an arbitrary point x, dropping terms of
O(h6). The order of term is removed by enclosing the taylor command with
convert( ,polynom). The quantity a is an integer which must be supplied.

> t:=a->y(x+a*h)=convert(taylor(y(x+a*h),h,6),polynom):

As a test, I. M. chooses a=2 and generates the Taylor series for y(x + 2h).

> t(2);

y(x + 2 h) = y(x) + 2 D(y)(x) h + 2 (D(2))(y)(x) h2 +
4
3

(D(3))(y)(x) h3

+
2
3

(D(4))(y)(x) h4 +
4
15

(D(5))(y)(x) h5

In the output, (D(4))(y)(x), for example, stands for the 4th derivative, y ′′′′(x).
Making use of the operator t, the proposed FDA is entered in eq1 .

> eq1:=-t(2)+16*t(1)-30*t(0)+16*t(-1)-t(-2);

eq1 := −y(x + 2 h) + 16 y(x + h) − 30 y(x) + 16 y(x − h) − y(x − 2 h)

= 12 (D(2))(y)(x) h2

On the rhs of the output, only the second derivative term survives, all other
terms up to O(h6) having canceled. Using the isolate command, the second
derivative (entered as D[1,1](y)(x)) is isolated to the lhs of eq2 ,

> eq2:=isolate(eq1,D[1,1](y)(x));

eq2 := (D(2))(y)(x)

= − 1
12

y(x + 2 h) − 16 y(x + h) + 30 y(x) − 16 y(x − h) + y(x − 2 h)
h2

thus confirming the FDA.
An arrow operator y is formed to evaluate y=x5.1 at an arbitrary x value.

So that the FDA can be evaluated for different h values, the rhs of eq2 is turned
into a functional operator A, depending on h, using the unapply command.

> y:=x->xˆ5.1: A:=unapply(rhs(eq2),h):

The exact y ′′(x) is calculated and the FDAs generated for h=0.1 and h=0.05.

> exact:=diff(y(x),x,x); approx1:=A(0.1): approx2:=A(0.05):

exact := 20.91 x3.1

The differences between the FDAs approx1 and approx2 and exact are plotted
for x=0.1 to 10 as solid red and dashed green curves, respectively, the resulting
plot being shown in Figure 2.2.

> plot([approx1-exact,approx2-exact],x=0.1..10,color=

[red,green],thickness=2,linestyle=[1,3],tickmarks=[2,2]);

–5e–05

0
5 10x

Figure 2.2: FDA minus the exact result for h=0.1 (solid) and h=0.05 (dashed).

As expected, the accuracy of the FDA is improved with the smaller value of h.
Both curves have greatest accuracy over a limited range of x.

Now, I. M. tackles part (b) of the problem. The plots library package is
loaded and 12 digits accuracy entered.

> restart: with(plots): Digits:=12:

The FDA involves a Taylor expansion in two variables. I. M. uses the multivari-
ate Taylor series command, mtaylor, to create an operator t for performing the
expansion of f(x + ah, y + bk) in powers of h and k, neglecting terms of sixth
order. mtaylor does not generate the order of term, so convert( ,polynom)
needn’t be applied. Two integers a and b must be entered as arguments in t.

> t:=(a,b)->f(x+a*h,y+b*k)=mtaylor(f(x+a*h,y+b*k),[h,k],6):

The given FDA is entered in eq1 .

> eq1:=t(1,1)+t(-1,1)+t(1,-1)+t(-1,-1)

-2*(t(1,0)+t(-1,0)+t(0,1)+t(0,-1))+4*t(0,0);

eq1 := f (x + h, y + k) + f (x − h, y + k) + f (x + h, y − k) + f (x − h, y − k)
− 2 f (x + h, y) − 2 f (x − h, y) − 2 f (x, y + k) − 2 f (x, y − k)
+ 4 f (x, y) = h2 k2 D1, 1, 2, 2(f)(x, y)

Only a single term involving D1, 1, 2, 2 results on the rhs of the output, the
subscripts 1 and 2 denoting derivatives with respect to x and y, respectively.
Dividing eq1 by h2k2, and simplifying, yields the desired FDA.

> eq2:=simplify(eq1/(hˆ2*kˆ2));

eq2 := (f(x + h, y + k) + f(x − h, y + k) + f(x + h, y − k) + f(x − h, y − k)
− 2 f(x + h, y) − 2 f(x − h, y) − 2 f(x, y + k) − 2 f(x, y − k)
+ 4 f(x, y))/(h2 k2) = D1, 1, 2, 2(f)(x, y)

The function f(x, y)=sin(x − y) e−x2 y2
is entered, as well as h=k=0.05.

> f:=(x,y)->sin(x-y)*exp(-xˆ2*yˆ2): h:=0.05: k:=0.05:

The exact 4th derivative is calculated and the FDA given by the lhs of eq2 .

> exact:=diff(f(x,y),x,x,y,y); approx:=lhs(eq2):

The plot3d command is used to create a 3-dimensional plot of approx minus
exact for x = −3...3, y = −3...3. The grid is taken to be 25 ×25, the shad-
ing to be XYZ (i.e., the color varies in the 3 directions), the axes boxed, the
plot “illuminated” with a light source at a certain orientation given by enter-
ing lightmodel=light2, the number of plotting points to be 2000, and the
orientation specified. The resulting picture is shown in Figure 2.3.

> plot3d(approx-exact,x=-3..3,y=-3..3,grid=[25,25],

shading=XYZ,axes=box,lightmodel=light2,numpoints=2000,
tickmarks=[3,3,3],orientation=[20,40]);

Figure 2.3: Difference between FDA and exact 4th derivative.

The regions where the FDA and exact result differ appreciably can be seen as
a series of ridges and valleys. The 3-dimensional plot can be rotated on the
computer screen by dragging with the mouse.

2.2 Series Solutions of LODEs

Consider an nth-order homogeneous linear ODE with variable coefficients,

dny

dxn
+ an−1(x)

dn−1y

dxn−1 + · · · + a1(x)
dy

dx
+ a0(x) y = 0. (2.4)

If a0(x), ..., an−1(x) are regular (single-valued and analytic) at a point x = x0,
then x0 is referred to as an ordinary point of the ODE. Near x0, the general
solution of the LODE can be written as a Taylor series, y=

∑∞
m=0 cm(x−x0)m,

whose radius of convergence is the distance to the nearest singular point (a
non-ordinary point) of the ODE. The coefficients cm are found by substituting
the series into the ODE and setting the coefficients of each power of x to zero.

If x0 is not an ordinary point, but (x − x0) an−1(x), (x − x0)2 an−2(x),...,
(x − x0)n a0(x) are regular at x0, then x0 is a regular singular point. Near such
an x0, a Frobenius series solution of the form y = (x−x0)s

∑∞
m=0 cm (x−x0)m,

with c0 �= 0 and s not necessarily an integer, can always be found. The radius
of convergence is again the distance to the nearest singular point outside x0.

In this section, series solutions to two familiar Sturm–Liouville ODEs are
obtained, illustrating the two types of expansions mentioned above.

2.2.1 Jennifer Renews an Old Acquaintance

We need two kinds of acquaintances, one to complain to,
while to the others we boast.
Logan Pearsall Smith, American essayist, Afterthoughts,“Other People”, 1931

In recipe 01-2-1, Jennifer introduced the Legendre functions. In this recipe,
she renews her acquaintance with these important special functions, showing
us how they arise as a series solution to the Legendre ODE,

(1 − x2) y ′′ − 2 x y ′ + n (n + 1) y = 0. (2.5)

Casting (2.5) into the “standard” form (2.4), Equation (2.5) has regular singular
points at x = ±1. Jennifer seeks a Taylor series solution about the ordinary
point x0 = 0. She decides that it will suffice for calculation purposes to work
with a finite number of terms, say N = 7, in the series.

> restart: N:=7:

The left-hand side of Legendre’s equation is now entered.
> ode:=(1-xˆ2)*diff(y(x),x,x)-2*x*diff(y(x),x)+n*(n+1)*y(x);

ode := (1 − x2) (
d2

dx2 y(x)) − 2 x (
d

dx
y(x)) + n (n + 1) y(x)

The power series y(x) =
∑N

m=0 cm xm is inputted using the add command.
> y(x):=add(c||m*xˆm,m=0..N);

y(x) := c0 + c1 x + c2 x2 + c3 x3 + c4 x4 + c5 x5 + c6 x6 + c7 x7

y(x) is automatically substituted into ode and powers of x are collected.

> ode2:=collect(ode,x);

ode2 := (−56 c7 + n (n + 1) c7 ) x7 + (−42 c6 + n (n + 1) c6 ) x6

+ (42 c7 − 30 c5 + n (n + 1) c5 ) x5 + (n (n + 1) c4 − 20 c4 + 30 c6 ) x4

+ (−12 c3 + n (n + 1) c3 + 20 c5 ) x3 + (−6 c2 + 12 c4 + n (n + 1) c2 ) x2

+ (6 c3 + n (n + 1) c1 − 2 c1 ) x + 2 c2 + n (n + 1) c0
Since ode2 represents the lhs of the Legendre equation, it must be set equal
to zero to form the complete equation. But x is arbitrary, so the coefficient of
each power of x must separately be equal to zero. A functional operator eq is
created to set the coefficient of xm in ode2 equal to zero.

> eq:=m->coeff(ode2,x,m)=0:

The set of equations eq(m) is solved for the c||(m+2) for m=0 to N − 2.

> sol:=solve({seq(eq(m),m=0..N-2)},{seq(c||(m+2),m=0..N-2)}):
The solution sol is assigned, the coefficients factored and the series formed,

> assign(sol): y:=add(factor(c||m)*xˆm,m=0..N):

and the coefficients of c0 and c1 collected in y.

> y:=collect(y,{c||0,c||1});

y := (1 − n (n + 1)x2

2
+

n (n − 2) (n + 3) (n + 1)x4

24

− n (n − 2) (n − 4) (n + 5) (n + 3) (n + 1)x6

720
)c0

+ (x − (n + 2) (n − 1) x3

6
+

(n − 1) (n − 3) (n + 4) (n + 2)x5

120

− (n − 1) (n − 3) (n − 5) (n + 6) (n + 4) (n + 2)x7

5040
)c1

y gives the first few term in the general series solution of Legendre’s ODE.
It consists of an even (coefficient of c0 ) and odd (coefficient of c1 ) series in
x. Noting that the denominator of the xm term is just m! (e.g., for m = 6,
6! = (6)(5)(4)(3)(2)(1) = 720), the structure of higher order terms in the infinite
series is easy to deduce. In deriving y , Jennifer has mimicked a hand calculation.
The same result could be more easily obtained by using the series option in
the dsolve command, as Jennifer will now demonstrate. She unassigns y and
sets the order of the first term to be neglected in the series solution. Since she
has taken N = 7, the order here is 8, i.e., terms of O(x8) are dropped in the
series. If the order is not specified, the default is to neglect terms of O(x6).

> unassign(’y’): Order:=N+1;

Order := 8
Then ode is solved for y(x), subject to the initial condition y(0) = c0 and
y ′(0)=c1 , using the series option in the dsolve command.

> Y:=dsolve({ode,y(0)=c||0,D(y)(0)=c||1},y(x),series);

The output (suppressed here in the text) contains the “order of” term O(x8)
which is then removed from the rhs of Y using convert( ,polynom).

> Y:=convert(rhs(Y),polynom):

Collecting the coefficients of c0 and c1 yields a series solution Y equivalent to
that obtained in y.

> Y:=collect(Y,{c||0,c||1});

Y := (1 − n (n + 1)x2

2
+ (

1
24

n4 +
1
12

n3 − 5
24

n2 − 1
4

n) x4

+ (− 1
720

n6 − 1
240

n5 +
23
720

n4 +
17
240

n3 − 47
360

n2 − 1
6

n) x6)c0

+ (x + (
1
3

− 1
6

n2 − 1
6

n) x3 + (− 13
120

n2 +
1

120
n4 +

1
60

n3 − 7
60

n +
1
5
) x5

+(
41

5040
n4 − 1

5040
n6 − 1

1680
n5 +

29
1680

n3 − 5
63

n2 − 37
420

n +
1
7
) x7)c1

Often, in cases of physical interest, x ≡ cos(θ) where the angle θ varies from
0 to π radians. So x then ranges from 1 to −1. It turns out that both series
diverge at x = ±1, unless n is a positive integer in which case the even or odd
series in Y , or y, terminates with the power xn when n is even or odd. To
explicitly demonstrate this, Jennifer turns Y into an operator depending on n
with the unapply command,

> YY:=unapply(Y,n):

and then uses a do loop to generate the solutions y0 ,y1 ,...,y5 for n=0, 1, ..., 5
> for n from 0 to 5 do y||n:=YY(n); end do;

y0 := c0 + (x +
1
3

x3 +
1
5

x5 +
1
7

x7) c1

y1 := (1 − x2 − 1
3

x4 − 1
5

x6) c0 + c1 x

y2 := (1 − 3 x2) c0 + (x − 2
3

x3 − 1
5

x5 − 4
35

x7) c1

y3 := (1 − 6 x2 + 3 x4 +
4
5

x6) c0 + (x − 5
3

x3) c1

y4 := (1 − 10 x2 +
35
3

x4) c0 + (x − 3 x3 +
6
5

x5 +
2
7

x7) c1

y5 := (1 − 15 x2 + 30 x4 − 10 x6) c0 + (x − 14
3

x3 +
21
5

x5) c1

In y0 , the coefficient of c0 is 1, which is just the zeroth order Legendre polyno-
mial P0(x), while the coefficient of c1 involves the leading terms of an infinite
series. In y1 , the c0 coefficient is an infinite series, while the c1 coefficient
is x, which is just the first order Legendre polynomial P1(x). In y2 , the c0
coefficient is proportional to P2(x), the coefficient of c1 an infinite series, and
so on for increasing n values. Jennifer creates a functional operator F to extract
the coefficient of either c0 (set p=0) or c1 (set p=1) for each solution yn.

> F:=(n,p)->coeff(y||n,c||p):

The Legendre polynomials Pn(x) are traditionally normalized so that each poly-
nomial has the value 1 at x = 1. Making use of F(2*n,0) and including this
normalization, the first few even subscript Legendre polynomials are generated.

> Peven:=seq(P||(2*n)=F(2*n,0)/eval(F(2*n,0),x=1),n=0..2);

Peven := P0 = 1, P2 = −1
2

+
3 x2

2
, P4 =

3
8

− 15
4

x2 +
35
8

x4

The odd subscript polynomials P1, P3, and P5 can be generated by replacing
F(2*n,0) with F(2*n+1,1) in the last command line.

The odd exponent infinite series are similarly extracted by using F(2*n,1).
These are labeled Q0 , Q2 , Q4 for reasons which will now be explained.

> Qeven:=seq(Q||(2*n)=F(2*n,1),n=0..2);

Qeven := Q0 = x +
1
3

x3 +
1
5

x5 +
1
7

x7, Q2 = x − 2
3

x3 − 1
5

x5 − 4
35

x7,

Q4 = x − 3 x3 +
6
5

x5 +
2
7

x7

Each of the above infinite series can be summed and are found to be proportional
to the Legendre functions Q0(x), Q2(x), Q4(x) of the second kind. Jennifer
concludes her recipe by demonstrating, for example, that the Taylor expansion
of Q0(x) is the same as Q0 given above. The solution branch between x=−1
and 1 is selected by entering the following command.

> _EnvLegendreCut:=1..infinity:

Then Taylor expanding LegendreQ(0,x) about x=0, dropping terms of O(x8),
and simplifying yields a series expansion for Q0 which agrees with that for Q0 .

> Q[0]:=simplify(LegendreQ(0,x)=taylor(LegendreQ(0,x),x=0,8));

Q0 :=
1
2

ln(x + 1) − 1
2

ln(1 − x) = x +
1
3

x3 +
1
5

x5 +
1
7

x7 + O(x8)

In physical problems where x varies from −1 to 1, the Qn(x) must be rejected
because they diverge at the end points of the range.

2.2.2 Another Old Acquaintance

Acquaintance. A person whom we know well enough to borrow from,
but not well enough to lend to.
Ambrose Bierce, American author, The Devils Dictionary, (1842–1914)

After revisiting the Legendre functions, it’s not too surprising that Jennifer
renews another old acquaintance, the Bessel functions, again showing how they
arise as a series solution to Bessel’s equation (with p non-negative),

x2 y ′′ + x y ′ + (x2 − p2) y = 0. (2.6)

Putting (2.6) into the standard form (2.4), the second-order LODE has a regular
singular point at x= x0 =0, so Jennifer will now seek a Frobenius power series
solution expanded about x=0, viz., y =

∑∞
m=0 cm xm+s. The total number N

of terms which will be kept in the series for plotting purposes is taken to be
100. The left-hand side of Bessel’s equation is entered in ode.

> restart: N:=100:

> ode:=xˆ2*diff(y(x),x,x)+x*diff(y(x),x)+xˆ2*y(x)-pˆ2*y(x);

ode := x2 (
d2

dx2 y(x)) + x (
d

dx
y(x)) + x2 y(x) − p2 y(x)

For y(x), Jennifer enters the summand of the Frobenius series.
> y(x):=c[m]*xˆ(m+s);

y(x) := cm x(m+s)

Noting that y(x) is automatically substituted into ode, ode is divided by xs and
the result simplified. The resulting summand is given by the output of eq .

> eq:=simplify(ode/xˆs);

eq := cm xm m2 + 2 cm xm m s + cm xm s2 + x(2+m) cm − p2 cm xm

The sum
∑∞

m=0 over the five terms of eq is to be set equal to zero which implies
that the coefficients of equal powers of x must also be equal to zero. Now all the
terms in eq involve xm, except for the fourth one which contains xm+2. But,
on summing, the fourth operand in eq can be transformed as follows,

∞∑
m=0

cm xm+2 =
∞∑

m=2

cm−2 xm =
∞∑

m=0

cm−2 xm. (2.7)

if we agree to define c−1 = 0, c−2 = 0. Jennifer enters these coefficient values.
> c[-1]:=0: c[-2]:=0:

The transformation (2.7) can be accomplished in eq by using the subsop com-
mand to replace the 4th operand with cm−2 xm.

> eq2:=subsop(4=c[m-2]*xˆm,eq);

eq2 := cm xm m2 + 2 cm xm m s + cm xm s2 + cm−2 xm − p2 cm xm

Now each term in eq2 involves xm. Dividing this equation by xm, equating the
result to 0, and simplifying, yields a recurrence relation relating cm to cm−2.

> eq3:=simplify(eq2/xˆm)=0;

eq3 := cm m2 + 2 cm m s + cm s2 + cm−2 − p2 cm = 0
Terms involving cm are now collected in the recurrence relation eq3 .

> eq4:=collect(eq3,c[m]);

eq4 := (m2 + 2 m s + s2 − p2) cm + cm−2 = 0
Setting m=0 in eq4 yields the so-called indicial equation, eq5 .

> eq5:=eval(eq4,m=0);

eq5 := (s2 − p2) c0 = 0
Assuming that the coefficient c0 �= 0, eq5 has two solutions for s, which are
explicitly extracted using the solve command. Jennifer will consider the series

solution (the second one in sol) to Bessel’s equation corresponding to s = p.
The other series solution is obtained by replacing p with −p.

> sol:=solve(eq5,s); s:=sol[2];

sol := −p, p

s := p

Evaluating the recurrence relation eq4 for m = 1, and noting that for non-
negative p, 1+2 p can never be zero, it follows from eq6 below that the coefficient
c1 =0. This coefficient is extracted from eq6 with the solve command.

> eq6:=eval(eq4,m=1); c[1]:=solve(eq6,c[1]);

eq6 := (1 + 2 p) c1 = 0
c1 := 0

For m ≥ 2, one must work with the full recurrence equation eq4 . The recurrence
relation can be rewritten by isolating the coefficient cm on the left side of the
equation and then factoring the result as in eq7 .

> eq7:=factor(isolate(eq4,c[m]));

eq7 := cm = − cm−2

m (m + 2 p)
The coefficients c2 to cN = c100 are now explicitly evaluated by first using
the unapply command to turn eq7 into a functional operator in terms of the
argument m and then using a do loop to iterate the recurrence relation.

> eq8:=unapply(eq7,m):

> for m from 2 to N do c[m]:=rhs(eq8(m)); end do:

Jennifer has used a command line ending colon to suppress the very long output.
All the even subscript coefficients are proportional to c0, while odd subscript
coefficients are proportional to c1 and therefore are all equal to zero.

To study the behavior of the series as more and more terms are retained, an
arrow operator JJ is introduced to add the terms in the series for m=0 to M .

> JJ:=M->add(c[m]*xˆ(m+s),m=0..M):

Jennifer now calculates the sum Jp for M =6 and collects the c0 coefficients.
> Jp:=collect(JJ(6),c[0]);

Jp :=
(
xp − x(2+p)

2 (2+2 p)
+

x(4+p)

8 (2+2 p) (4+2 p)
− x(6+p)

48 (2+2 p) (4+2 p) (6+2 p)

)
c0

With the proper assignment (which will be done shortly) of the arbitrary coef-
ficient c0, the infinite series corresponding to Jp is the Bessel function Jp of the
first kind of order p. Not surprisingly, the same result could have been much
more easily obtained by using the series option of the dsolve command. To
demonstrate this, Jennifer now unassigns y(x). The order of the series is set as
well as the information level on the methods used in the dsolve command.

> unassign(’y(x)’): Order:=7: infolevel[dsolve]:=5:

Jennifer replaces y(x) with z(x) in ode and obtains the general series solution
of ode2. She has deleted some of the unsuccessful methods in the output.

> ode2:=subs(y(x)=z(x),ode): dsolve(ode2,z(x),series);

....................................
dsolve/series/ordinary: trying Newton iteration
dsolve/series/direct: trying direct subs
dsolve/series/froben: trying method of Frobenius
dsolve/series/froben: indicial eqn is -pˆ 2+rˆ 2
dsolve/series/froben: roots of indicial eqn are [[p], [-p]]

z (x) = C1 xp(1 − 1
4 p + 4

x2 +
1

(8 p + 16) (4 p + 4)
x4

− 1
(12 p + 36) (8 p + 16) (4 p + 4)

x6 + O(x7))

+ C2 x(−p)(1 +
1

4 p − 4
x2 +

1
(8 p − 16) (4 p − 4)

x4

+
1

(12 p − 36) (8 p − 16) (4 p − 4)
x6 + O(x7))

The method of Frobenius has been successfully used, the roots of the indicial
equation obtained, and the general series solution constructed for z(x). The
first series, involving the arbitrary coefficient C1, is exactly the same as in Jp.
The second series, involving the other arbitrary coefficient C2, is the second
independent solution of the second order LODE obtained by replacing p with
−p. I.e., the second independent solution is J−p(x).

This conclusion about the mathematical form of the second solution is true
providing that p is not zero or an integer. For p = 0, there is only one root,
s = 0, to the indicial equation and the second series is identical with the first.
For integer values of p, all denominators in the second series beyond a certain
term vanish, so this series becomes meaningless. Thus for p=0, 1, 2, ... a more
general second solution must be sought. Jennifer has decided not to go into
this issue here, referring her students to standard ODE texts which cover the
topic. At this stage it suffices to note that this second independent solution is
the Bessel function Yp of the second kind of order p.

Returning to her “hand mimicking” calculation, Jennifer completes the iden-
tification of the infinite series solution as Jp(x) by making the “standard” choice
for c0, viz. c0 = 1/(2p p!).

> c[0]:=1/(2ˆp*p!):

To see how may terms have to be retained in the series over a certain range
of x, e.g., x = 0 to 50, to obtain a reasonably correct curve, Jennifer will now
plot the sequence of results obtained from JJ for p = 0 and N/5 = 100/5 = 20,
40, 60, 80, and 100 terms along with the “exact” infinite series result for J0(x).
Because the finite series results diverge to ∞ before x = 50 is reached, the
vertical view is limited to be between −1 and 1.

> plot([seq(eval(JJ(k*N/5),p=0),k=1..5),BesselJ(0,x)],

x=0..50,thickness=2,numpoints=500,labels=["x","J"],
tickmarks=[3,3],view=[0..50,-1..1]);

Figure 2.4: Divergent curves from left to right: 20, 40, 60, 80, and 100 terms.

From left to right in Figure 2.4, the divergent curves represent the finite series
representations of J0 for 20, 40, 60, 80, and 100 terms. The non-divergent
oscillatory curve is the infinite series representing the exact J0(x).

2.3 Fourier Series

Consider a single-valued function f(x) defined over the fundamental interval
−L ≤ x ≤ L and satisfying the boundary conditions f(−L) = f(L). If f(x) has
a finite number of discontinuities and maxima and minima, and

∫ L

−L
|f(x)| dx

is finite,1 then f(x) can be expanded in the Fourier series ([MW71], [Boa83])

f(x) =
1
2
a0 +

∞∑
n=1

[
an cos

(nπx

L

)
+ bn sin

(nπx

L

)]
, (2.8)

with an =
1
L

∫ L

−L

f(x) cos
(nπx

L

)
dx, bn =

1
L

∫ L

−L

f(x) sin
(nπx

L

)
dx.

The forms of an and bn can be derived from Equation (2.8) by noting that
yn(x) = cos(n π x/L) (or sin(n π x/L)) satisfies the orthogonality condition∫ L

−L
w(x) ym(x) yn(x) dx = 0, for m �= n, with w(x) = 1. The identification of

w(x) follows on noting that the yn(x) are solutions of y ′′(x)+(nπ/L)2y(x)=0.
This equation is a Sturm–Liouville ODE, (1.6), with p = 1, q = 0, w = 1, and
λ=−(nπ/L)2. The an and bn follow on multiplying (2.8) by cos(n π x/L) (or
sin(n π x/L)), integrating from −L to L, and using the orthogonality condition.

If f(x) is an odd function, that is to say f(−x) = −f(x), then an = 0 and
bn =(2/L)

∫ L

0 f(x) sin(n π x/L) dx, so f(x) is expressed as a Fourier sine series.

1These are sufficient, but not necessary, conditions.

On the other hand, if f(x) is an even function, i.e., f(−x) = f(x), then
an =(2/L)

∫ L

0 f(x) cos(n π x/L) dx and bn =0, so f(x) is a Fourier cosine series.
Since each term in (2.8) is periodic with period 2L, then f(x + 2L)=f(x).

Thus, the Fourier series may either represent an f(x) defined in the fundamental
interval (−L, L), or a periodic f(x) with period 2L for all of x.

When f(x) is defined only in the range 0 to L, it can be written as a Fourier
sine series by including the range to −L to 0 and considering f(x) to be an odd
function about x = 0. Alternately, it can be written as a Fourier cosine series
by considering f(x) to be an even function about the origin. It may turn out
that one series fits f(x) better than the other for a finite number of terms.

To this point, the fundamental interval has been taken to be 2L. This can
be easily changed. For example, consider f(t) defined in the range t = 0 to T
and we want the fundamental interval to be T , not 2T . To accomplish this, set
x = t and L = T/2 in (2.8) and an and bn and change the range of the integrals
from −T/2 ... T/2 to 0 ... T . In this case, the general Fourier expansion becomes

f(t) =
1
2
a0 +

∞∑
n=1

[
an cos

(
2nπt

T

)
+ bn sin

(
2nπt

T

)]
, (2.9)

with an =
2
T

∫ T

0
f(t) cos

(
2nπt

T

)
dt, and bn =

2
T

∫ T

0
f(t) sin

(
2nπt

T

)
dt.

Again, this series may be used to represent either a function defined in the
fundamental interval 0 to T or a periodic function whose period is T .

The concept of expanding a function f(x) in terms of sines and cosines can
be extended to other special functions yn(x) satisfying a S-L type equation. If
the yn(x) satisfy the same boundary conditions at a and b as f(x), then

f(x) =
∑

n

An yn(x), with An =

∫ b

a
w(x) f(x) yn(x) dx∫ b

a
w(x) yn(x)2 dx

. (2.10)

The functions yn(x) are said to form a complete set. Often, they are normalized
so that

∫ b

a
w(x) yn(x)2 dx=1. Since they have the orthogonality property, they

then satisfy the orthonormality condition∫ b

a

w(x) ym(x) yn(x) dx = δmn, (2.11)

where δmn, the Kronecker delta, is defined by δmn =1 for m=n and 0 for m �=n.
As an example of expanding in terms of special functions, the Legendre–

Fourier series (or simply the Legendre series) is given by

f(x) =
∞∑

n=0

An Pn(x), with An =
(2 n + 1)

2

∫ 1

−1
f(x) Pn(x) dx. (2.12)

A mathematical example of this series is given in Recipe 02-3-3.

2.3.1 Madeiran Levadas and the Gibb’s Phenomenon

The idealist walks on tiptoe, the materialist on his heels.
Malcolm de Chazal, French writer, (1902–81)

On the island of Madeira, water is transported from the mountains by a net-
work of levadas (irrigation canals) which often cling to the mountain side with
vertigo-inducing drop offs and pass through pitch-black tunnels. This recipe
is inspired by some interesting hikes that I have taken on Madeiran levada re-
taining walls. A levada retaining wall is described by the piecewise function
f = (L + x)/2 for −L ≤ x ≤ −L/2, f = L/4 for −L/2 ≤ x < 0, and f = 0 for
0 < x ≤ L, with L=π. Determine the Fourier series representation of f(x) and
plot it and f together. Calculate the cross-sectional area of the retaining wall
using f(x) and then the Fourier series. Discuss the various results.

To simplify the command entries, let’s set X =π x/L.
> restart: X:=Pi*x/L:

Using the inert Sum command, an operator F is formed to calculate the Fourier
series, keeping N terms.

> F:=N->a[0]/2+Sum(a[n]*cos(n*X)+b[n]*sin(n*X),n=1..N);

F := N → 1
2

a0 + (
N∑

n=1

(an cos(nX) + bn sin(nX)))

The formal expressions for the coefficients are entered using the inert form of
the integral command. In the outputs, X is replaced with π x/L.

> a[0]:=(1/L)*Int(f,x=-L..L);

a0 :=
1
L

∫ L

−L

f dx

> a[n]:=(1/L)*Int(f*cos(n*X),x=-L..L);

an :=
1
L

∫ L

−L

f cos(
nπ x

L
) dx

> b[n]:=(1/L)*Int(f*sin(n*X),x=-L..L);

bn :=
1
L

∫ L

−L

f sin(
nπ x

L
) dx

The value L=π is specified and the piecewise function f entered.
> L:=Pi: f:=piecewise(x<-L/2,(L+x)/2,x<0,L/4,x<L,0);

f :=

⎧⎪⎪⎨
⎪⎪⎩

π

2
+

x

2
x < −π

2
π

4
x < 0

0 x < π
The value of the coefficient a0 is obtained.

> a[0]:=value(a[0]);

a0 :=
3 π

16
The coefficients an and bn can be simplified by assuming that n is an integer.
The “type match command” (::) is used in the assumption.

> a[n]:=simplify(value(a[n])) assuming n::integer;

an := −1
2

(−1)n − cos(
π n

2
)

π n2

> b[n]:=simplify(value(b[n])) assuming n::integer;

bn := −1
4

π n + 2 sin(
π n

2
)

π n2

The Fourier series is generated in FF , with N =15. (Only the leading terms in
the output are shown here in the text.) You can increase the value of N , but
you might then wish to suppress the very lengthy output by putting a colon on
the end of the command line.

> FF:=value(F(15));

FF :=
3 π

32
+

1
2

cos(x)
π

− 1
4

(π + 2) sin(x)
π

− 1
4

cos(2 x)
π

− 1
8

sin(2x)

+
1
18

cos(3 x)
π

− 1
36

(3 π − 2) sin(3x)
π

− 1
16

sin(4x) +
1
50

cos(5 x)
π

+ · · ·

Finally, the function f and the Fourier series FF are plotted over the range
x=−L=−π to x=L=π, being represented by thick blue and red lines. The
resulting picture is shown on the left of Figure 2.5.

> plot([f,FF],x=-L..L,color=[blue,red],thickness=2,axes=box);

–3 –2 –1 0 1 2 3x

Figure 2.5: Left: Fourier series for N =15 and f . Right: Series for N =100.

The Fourier series oscillates around the exact f . The fit can be improved (the
size of the oscillations reduced) by increasing N . The plot on the right of
Figure 2.5 shows the Fourier series result for N = 100. The overshoot in the
vicinity of the step function at x=0 persists, however, no matter how large an
N is chosen. This is called the Gibbs’ phenomenon. Notice also in the left plot
that the Fourier series curve passes approximately through the midpoint of the
step. As N → ∞, the Fourier curve will pass exactly though the midpoint, a
general property of Fourier series at step discontinuities.

The exact cross-sectional area of the retaining wall is calculated in Area1 by
integrating f from x=−L=−π to π. The cross-sectional area is also calculated
in Area2 by integrating the Fourier series FF over the same range.

> Area1:=int(f,x=-Pi..Pi); Area2:=int(FF,x=-Pi..Pi);

Area1 :=
3 π2

16
Area2 :=

3 π2

16
The two areas are identical. The “wiggles” in the Fourier series about the f
curve exactly cancel. You might like to confirm that this is still true for N =100.

2.3.2 Sine or Cosine Series?

I see it all perfectly; there are two possible situations – one can either
do this or that. My honest opinion and my friendly advice is this:
do it or do not do it – you will regret both.
Søren Kierkegaard, Danish philosopher, (1813–55)

Consider the function f(x) = x for 0 ≤ x ≤ L/2 and f(x) = L − x for
L/2 ≤ x ≤ L, with L = 1. Extending the range to −L ≤ x ≤ L, derive a
Fourier sine series and a cosine series representation of f(x). Plot f(x) and the
two series together and discuss the results.

The value of L is entered and, again for convenience, let’s set X =π x/L.
> restart: L:=1: X:=Pi*x/L:

The function f is entered using the piecewise command.
> f:=piecewise(x<L/2,x,x<L,L-x);

f :=

{
x x <

1
2

1 − x x < 1
An odd function f1 is introduced which is the same as f for 0 ≤ x ≤ L, but is
equal to −f for −L ≤ x ≤ 0. f1 will be used to generate the sine series.

> f1:=piecewise(x<-L/2,-(L+x),x<L/2,x,x<L,L-x);

f1 :=

⎧⎪⎪⎪⎨
⎪⎪⎪⎩

−1 − x x <
−1
2

x x <
1
2

1 − x x < 1

An even function f2 is introduced which is the same as f for 0 ≤ x ≤ L and
symmetrical about the origin. f2 is used to generate the cosine series.

> f2:=piecewise(x<-L/2,L+x,x<0,-x,x<L/2,x,x<L,L-x);

To confirm that f1 and f2 are odd and even extensions of f to the region x < 0,
they are plotted as solid and dashed curves in Figure 2.6. For x > 0, the two
curves are identical, and therefore indistinguishable.

> plot([f1,f2],x=-L..L,linestyle=[SOLID,DASH],thickness=2);

–0.4

0.4

–1 1
x

Figure 2.6: Solid curve, f1. dashed curve, f2.

For a given f , functional operators A, B and F are introduced to calculate the
series coefficients an and bn, and to produce the series out to N terms.

> A:=(f,n)->(1/L)*int(f*cos(n*X),x=-L..L);

A := (f, n) → 1
L

∫ L

−L

f cos(nX) dx

> B:=(f,n)->(1/L)*int(f*sin(n*X),x=-L..L);

B := (f, n) → 1
L

∫ L

−L

f sin(nX) dx

> F:=(f,N)->A(f,0)/2+sum(A(f,n)*cos(n*X)+B(f,n)*sin(n*X),n=1..N);

F := (f, N) → 1
2

A(f, 0) + (
N∑

n=1

(A(f, n) cos(nX) + B(f, n) sin(nX)))

Taking N =10, the Fourier sine and cosine series are produced in F1 and F2 .
> F1:=F(f1,10); F2:=F(f2,10);

F1:=
4 sin(π x)

π2 − 4
9

sin(3π x)
π2 +

4
25

sin(5π x)
π2 − 4

49
sin(7π x)

π2 +
4
81

sin(9π x)
π2

F2:=
1
4

− 2 cos(2 π x)
π2 − 2

9
cos(6 π x)

π2 − 2
25

cos(10 π x)
π2

The two series are plotted along with f and shown on the left of Figure 2.7.
> plot([f,F1,F2],x=0..L,color=[blue,red,green],thickness=2);

0

0.5

x 1

–0.01

0

0.01

0.02

1x

Figure 2.7: Left: f and the two series. Right: Solid, f − F1 . Dashed, f − F2 .

Even though N is not large, both series fit f quite well, except near the apex
of the triangle and, for the cosine series, near x = 0 and 1. To magnify the
difference between the series results and f , the differences f − F1 and f − F2
are plotted as solid and dashed curves in the right graph of Figure 2.5.

> plot([f-F1,f-F2],x=0..L,linestyle=[1,3],thickness=2);

The cosine series clearly fits less well at the end points of the x range. Can you
suggest why this is the case?

2.3.3 How Sweet This Is!

Few things are harder to put up with than the annoyance
of a good example.
Mark Twain, American author, Pudd’nhead Wilson, 1894

If you have calculated series expansions by hand, you know what a tedious
task it can be to explicitly calculate a large number of terms in the series and
then have to plot the results. By now, you should have gotten a clear idea that
using computer algebra is the way to go in handling such problems. The fol-
lowing recipe for the Legendre–Fourier series is a particularly “sweet” example
that derives a “beautiful” result very quickly.

Consider a step function, f(x)=0 for −1 < x < 0 and f(x)=1 for 0 < x < 1.
Derive the Legendre–Fourier series for f and plot the series and f together over
the range −1 < x < 1. Calculate the area between the x-axis and the series
curve and compare with the exact result for f . Discuss the plot and area results.

> restart:

Functional operators A and F are formed to calculate the coefficients An, and
the Legendre series out to N terms, for a given function f .

> A:=(f,n)->((2*n+1)/2)*int(f*LegendreP(n,x),x=-1..1);

A := (f, n) → 1
2

(2 n + 1)
∫ 1

−1
f LegendreP(n, x) dx

> F:=(f,N)->sum(A(f,n)*LegendreP(n,x),n=0..N);

F := (f, N) →
N∑

n=0

A(f, n) LegendreP(n, x)

The given f is entered with the Heaviside(x) command and the Legendre-
Fourier series calculated in F1 for N =15.

> f:=Heaviside(x); F1:=F(f,15);

f := Heaviside(x)

F1 :=
1
2

+
3 x

4
− 7

16
LegendreP(3, x) +

11
32

LegendreP(5, x)

− 75
256

LegendreP(7, x) +
133
512

LegendreP(9, x) − 483
2048

LegendreP(11, x)

+
891
4096

LegendreP(13, x) − 13299
65536

LegendreP(15, x)

This is a formidable looking series, which has been generated in three command
lines. If I now wanted to see what the series looks like for, say N =100, changing
15 to 100 in F1 and executing the command would generate the new result
almost instantaneously.

How formidable the result really is, even for 15 terms, can be appreciated by
expanding F1 . The sort command is used to order the polynomial expansion
from the highest exponent to the lowest. Wow, what a result!

> F1:=sort(expand(F1));

F1 := −128931743655
134217728

x15 +
503889568875

134217728
x13 − 800852700375

134217728
x11

+
664630841875

134217728
x9 − 307629132525

134217728
x7 +

78646056489
134217728

x5

− 10402917525
134217728

x3 +
703956825
134217728

x +
1
2

Now f and F1 are plotted in Figure 2.8 over the range x = −1 to 1, and are
represented by dashed and solid curves, respectively.

> plot([f,F1],x=-1..1,thickness=2,linestyle=[3,1]);

Two features of the Legendre series are quite clear from the picture, and can be
confirmed by taking N larger. (You will have to adjust the view and number
of digits and plotting points.) The series curve displays a Gibb’s phenomenon
and also passes through the midpoint of the step.

0

1

–1 1x

Figure 2.8: Dashed curve, Step function f . Solid curve, Legendre series F1 .

The area between the Legendre series curve and the x-axis is now calculated
and found to be exactly the same as the area for the step function.

> Area:=int(F1,x=-1..1);

Area := 1

2.4 Summing Series

In this section, two different approaches to summing infinite series are presented.
The first recipe makes use of ideas already introduced in the earlier Fourier
series examples. In the second recipe, the series to be summed is replaced by a
complex series which Maple is able to sum.

2.4.1 I. M. Curious Sums a Series

Once, I thought I made a mistake, but I was mistaken.
From the diary of I. M. Curious

In this recipe, Ms. Curious answers the following question:
By expanding f(x) = x(L − x), defined in the interval (0, L) with L=π, in

a Fourier sine series of period 2L and setting x = L/2, prove that
∞∑

n=0

(−1)n

(2n + 1)3
=

π3

32
.

Confirm this result by directly summing the series with Maple, first showing
that the sum can be expressed as either a generalized hypergeometric function
or as a polylogarithm function.

I. M. begins her solution by assuming that the Fourier series summation
indices m and n are integers. She then sets L=π and enters f =x (L − x).

> restart: assume(m::integer,n::integer):

> L:=Pi: f:=x*(L-x):

To expand f in a Fourier sine series of period 2L, she forms the following odd
piecewise function, pw , defined in the interval −L < x < L.

> pw:=piecewise(x<0,x*(L+x),x>0,f);

pw :=
{

x (π + x) x < 0
x (π − x) 0 < x

I. M. then calculates the Fourier coefficients a0, am for m �= 0, and bm.
> a[0]:=(1/L)*int(pw,x=-L..L);

a0 := 0

> a[m]:=(1/L)*int(pw*cos(m*Pi*x/L),x=-L..L);

am := 0

> b[m]:=(1/L)*int(pw*sin(m*Pi*x/L),x=-L..L);

bm := −4 (−1 + (−1)m)
π m3

She notices that in bm the coefficients are only non-zero if m is an odd integer.
So she substitutes m = 2n + 1 into bm and relabels the coefficients as b2n+1.
The new summation index n will take on the values n = 0, 1, 2, . . ..

> b[2*n+1]:=simplify(subs(m=2*n+1,b[m]));

b2 n+1 :=
8

π (2 n + 1)3

Since the coefficients am are zero for all m (and therefore all n), the Fourier
series is then of the form F =

∑∞
n=0 b2n+1 sin((2n + 1)πx/L).

> F:=Sum(b[2*n+1]*sin((2*n+1)*Pi*x/L),n=0..infinity);

F :=
∞∑

n=0

(
8 sin((2n + 1)x)

π (2 n + 1)3
)

Thus, the original function f in the region 0 < x < L can be written as the
Fourier sine series F . This is entered in eq1 .

> eq1:=f=F;

eq1 := x (π − x) =
∞∑

n=0

(
8 sin((2n + 1)x)

π (2 n + 1)3
)

eq1 is divided by 8 and evaluated at x=L/2.
> eq2:=eval(eq1/8,x=L/2);

eq2 :=
π2

32
=

1
8

(∞∑
n=0

(
8 (−1)n

π (2 n + 1)3
)

)

Multiplying eq2 by π, and expanding, confirms the series sum.

> eq3:=expand(Pi*eq2);

eq3 :=
π3

32
=

∞∑
n=0

(−1)n

(2 n + 1)3

To sum the series directly with Maple, I. M. extracts it from the rhs of eq3 .

> S:=rhs(eq3);

S :=
∞∑

n=0

(−1)n

(2 n + 1)3

She expresses the sum S as a hypergeometric function by applying the following
convert command.

> S:=convert(S,hypergeom);

S := hypergeom([
1
2
,

1
2
,

1
2
, 1], [

3
2
,

3
2
,

3
2
], −1)

If this function is unfamiliar to you, highlight hypergeom in the computer out-
put with your mouse and open the relevant help window to see its definition.
According to Help, it may be possible to convert the hypergeometric function
into one of the standard special and elementary functions found in such texts
as Handbook of Mathematical Functions by Abramowitz and Stegun ([AS72]).
Applying the convert(StandardFunctions) command,

> S:=convert(S,StandardFunctions);

S := −1
2

I polylog(3, I) +
1
2

I polylog(3, −I)

yields a combination of polylog functions. Again, if the polylogarithm function
is unfamiliar, it may be looked up in Maple’s Help. I. M. finally obtains the
desired form of the series sum by using simplify.

> S:=simplify(S);

S :=
π3

32

2.4.2 Spiegel’s Series Problem

Old age is that time of life when you can feel bad in the morning
without having had fun the night before.
Gregarius Nerd, Professor of Mathematics, Erehwon Institute of Technology

In the previous example, we saw that Maple was successful in summing the
given series. If Maple is unsuccessful, does that mean that the series cannot be
summed? Not necessarily, as you will now see. Sometimes it needs a bit of help
in the form of human brain power. This will not be the last time in this book
that this is the case.

Let’s consider the following infinite series,

r sin(φ) +
1
3

r3 sin(3φ) +
1
5

r5 sin(5φ) +
1
7

r7 sin(7φ) + · · · , (2.13)

which, according to the Schaum Outline Series on Advanced Mathematics by
Murray Spiegel ([Spi71]), arises from solving for the steady-state temperature

distribution in a thin circular plate of unit radius whose faces are insulated and
has each half of its boundary kept at a different constant temperature. In polar
coordinates, r is the radial distance from the center of the plate and φ the polar
angle. Spiegel’s Problem 12.52 is to show that the series can be summed and
cast into the form (1/2) tan−1(2r sin φ/(1 − r2)).

The following recipe solves Spiegel’s problem. An operator S is formed to
generate the series out to exponent 2 N + 1.

> restart:

> S:=N->sum(rˆ(2*m+1)*sin((2*m+1)*phi)/(2*m+1),m=0..N);

S := N →
N∑

m=0

r(2 m+1) sin((2m + 1)φ)
2 m + 1

Taking N =3, then entering S(3) generates the terms displayed in (2.13).
> S1:=S(3);

S1 := r sin(φ) +
1
3

r3 sin(3φ) +
1
5

r5 sin(5φ) +
1
7

r7 sin(7φ)

Taking N = ∞ in S and applying the value command, we find that Maple is
unable to directly sum the series, returning the unevaluated sum in S2 .

> S2:=value(S(infinity));

S2 :=
∞∑

m=0

r(2 m+1) sin((2m + 1)φ)
2 m + 1

To sum the series, note that the real series can be written as the imaginary part
of a complex series. To accomplish this, the summand in S2 can be taken as
the imaginary part of a complex summand, viz., with I ≡ √−1,

Im
(

r2m+1eI(2m+1)φ

2m + 1

)
=Im

(
(reIφ)2m+1

2m + 1

)
=Im

(
z2m+1

2m + 1

)

with z ≡ reIφ. Using this result, the complex series, CS , is entered and then
successfully summed.

> CS:=Sum(zˆ(2*m+1)/(2*m+1),m=0..infinity);

CS :=
∞∑

m=0

z(2 m+1)

2 m + 1
> CS:=value(CS);

CS :=
1
2

ln(
1 + z

1 − z
)

Then, z=reIφ is substituted into CS ,
> CS2:=subs(z=r*exp(I*phi),CS);

CS2 :=
1
2

ln(
1 + r e(φ I)

1 − r e(φ I) )

and the complex result CS2 broken into real and imaginary parts in CS3 with
the complex evaluation command and simplified in CS4 assuming r < 1.

> CS3:=evalc(CS2);

> CS4:=simplify(CS3) assuming r<1;

CS4 :=
1
4

ln(
r2 + 2 r cos(φ) + 1
1 − 2 r cos(φ) + r2 )

+
1
2

I arctan(
2 r sin(φ)

1 − 2 r cos(φ) + r2 , − −1 + r2

1 − 2 r cos(φ) + r2 )

The portion of the argument in arctan before the comma is the numerator, the
portion after the comma being the denominator. To extract the imaginary part
of CS4 , the coeff command is used to pull out the coefficient of I.

> S3:=coeff(CS4,I);

S3 :=
1
2

arctan(
2 r sin(φ)

1 − 2 r cos(φ) + r2 , − −1 + r2

1 − 2 r cos(φ) + r2 )

The operand command, op is used to write the series sum in a form which
agrees with the result quoted by Spiegel.

> S4:=op(1,S3)*arctan((op([2,1],S3)/op([2,2],S3)));

S4 := −1
2

arctan(
2 r sin(φ)
−1 + r2 )

2.5 Supplementary Recipes
02-S01: Euler and Bernoulli Numbers
(a) Taylor expand sec(z), dropping terms of O(z12). The Euler numbers E2n

are defined by sec(z)=
∑∞

n=0(−1)n E2n z2n/(2n)! Using the Euler number
command euler(2*n), confirm that the latter expansion of sec(z) agrees
with the Taylor expansion. Generate the Euler numbers E0, E2, E4,...,E10.

(b) Taylor expand z/(ez − 1), dropping terms of O(z12). The Bernoulli num-
bers Bn are defined by z/(ez − 1) =

∑∞
n=0 Bn zn/n! Using the Bernoulli

number command bernoulli(n), confirm that the latter expansion agrees
with the Taylor expansion. Generate the first 10 Bernoulli numbers.

02-S02: Ms. Curious Approximates an Integral
Ms. Curious has been given the following problem to solve. Consider the integral
f(x)=

∫ x

0 sin(t2) dt. Evaluate this integral analytically and identify the function
which occurs. Obtain the smallest polynomial approximation to f(x) which is
valid within ±0.00001 for 0 ≤ x ≤ 1.

02-S03: More Finite Difference Approximations
(a) Confirm the following finite difference approximation,

y ′′′′(x) = [y(x + 2h) − 4y(x + h) + 6y(x) − 4y(x − h) + y(x − 2h)]/h4.

Suggest a physical problem for which this FDA might be useful. Taking
y = x4.7, graphically compare and discuss the FDA (with h = 0.1 and 12
digits accuracy) with the exact 4th derivative over the range x=0 to 70.

(b) Show that an FDA to the Laplacian, ∇2f(x, y)≡∂2f/∂x2+∂2f/∂y2, is

∇2f(x, y)=(1/12 h2)[16 [f(x+h, y) + f(x, y+h)+f(x − h, y)+f(x, y − h)]

−[f(x+2h, y)+f(x, y+2h)+f(x − 2 h, y)+f(x, y − 2 h)+60 f(x, y)]].

Suggest a physical problem for which this FDA might be useful. Consid-
ering f(x, y)=e−x2 y2

and taking h=0.1 and 10 digits accuracy, produce
a 3-dimensional color-coded plot of the difference between the exact 2-
dimensional Laplacian and the FDA for the range x = −2..2, y = −2..2.

02-S04: Series Solution
Mimicking a hand calculation, obtain a general series solution, valid near x=0,
of the following LODE

x y ′′ + 2 y ′ + x y = 0.
Show that the series solution may be expressed in a closed form. Confirm this
closed-form solution by directly solving the LODE with the dsolve command.

02-S05: Chebyshev Polynomials Revisited
Obtain a general series solution of Chebyshev’s equation,

(1 − x2) y ′′ − x y ′ + p2 y = 0
valid near x = 0, by (a) mimicking a hand calculation, (b) using the series
option in the dsolve command. If p is a positive integer, show that one or
the other of the series in the general solution reduces to a finite polynomial.
These polynomials are the Chebyshev polynomials. The zeroth order Chebyshev
polynomial T0(x) = 1. The higher order polynomials Tm(x) are normalized so
that the coefficient of the largest power in the mth order polynomial is 2(m−1).
Derive the Chebyshev polynomials T1, T2, ...,T7 and plot T1 to T5 over the
range x = −1 to 1. Note that the non-finite-polynomial parts of the solution
diverge at the end points of the range and are rejected in physical problems.

02-S06: A Fourier Series
Expand f(θ)=θ2, 0 < θ < 2 π, in a Fourier series F of period 2π. Plot f and
F (for an upper value N =20 of the summation index) in the same figure.

02-S07: Fourier Sine Series
Taking L = 1, expand each of the following f(x) in a Fourier sine series F of
period 2L, over the interval (0,L):

(a) f(x)=x (L − x);

(b) f(x)=x, 0<x<L/2, and f(x)=L − x, L/2<x<L;

(c) f(x)=1, 0<x<L/2, and f(x)=0, L/2<x<L.

In each case plot f(x) and F (for an upper value N = 10 of the summation
index) in the same figure and discuss the goodness of the fit.

02-S08: Fourier Cosine Series
Taking L=1, expand each of the f(x) given in 02-S07 in a Fourier cosine series

F of period 2L, over the interval (0,L). In each case plot f(x) and F (for an
upper value N =10 of the summation index) in the same figure and discuss the
goodness of the fit.

02-S09: Legendre Series
Expand the following f(x) in a series F of Legendre polynomials (up to n =
N =15) and plot f and F together in the same figure:

(a) f(x)=0, −1<x<0, and f(x)=1, 0<x<1;

(b) f(x)=0, −1<x<0, and f(x)=x, 0<x<1.

02-S10: Directly Evaluating Series Sum
Write the following infinite series out in the summation notation and then use
Maple to directly evaluate the series sum in closed form:

(a) f(x) = 1 + 2x + 3 x2 + 4 x3 + · · ·

(b) f(x) =
1

1 · 2
+

x

2 · 3
+

x2

3 · 4
+

x3

4 · 5
+ · · ·

(c) f =
1

1 · 3
+

1
2 · 4

+
1

3 · 5
+

1
4 · 6

+ · · ·

(d) f(x) = x − 4 x3

3!
+

9 x5

5!
− 16 x7

7!
+ · · ·

02-S11: Another Cosine Series
Expand f(x) = sin(x), 0 < x < π, in a Fourier cosine series F . Explicitly write
out F for the first 5 non-zero terms and plot F and f(x) together. Use F to
prove

∑∞
n=1 1/(4n2 − 1)=1/2. Confirm this by directly summing with Maple.

02-S12: The Complex Series Trick Again
In a certain 2-dimensional electrostatic potential problem, the following infinite
series occurs:

S = sin(
π x

a
) e−

π y
a +

1
3

sin(
3 π x

a
) e−

3 π y
a +

1
5

sin(
5 π x

a
) e−

5 π y
a + · · ·

Show that the series can be summed and put into the form

S =
1
2

arctan
(

sin(π x/a)
sinh(π y/a)

)
.



Chapter 3

Vectors and Matrices

In the first section of this chapter, we see how Maple may be used to deal with
vectors in Cartesian coordinates. Examples of vector algebra, the dot and cross
products, the gradient operator, and vector identities are presented.

The second section extends the discussion to vectors in orthogonal1 curvi-
linear coordinate systems such as spherical polar, cylindrical, and others. The
vector operators gradient, divergence, curl, and Laplacian are considered and
various important identities and theorems are illustrated.

The third section looks at the manipulation of matrices. Examples of matrix
addition and multiplication, calculating the transpose and inverse and eigen-
values and eigenvectors, and diagonalizing and rotating matrices, are provided.

3.1 Vectors: Cartesian Coordinates

For Cartesian coordinates x, y, and z, the unit vectors êx, êy, and êz always
point along the x, y, and z axes for every point in space. A general vector �A
is of the form �A=Ax êx + Ay êy + Az êz. If �A is a function of the coordinates,
then �A is called a vector field.

The sum of two vectors �A and �B is given by
�A + �B=(Ax + Bx) êx + (Ay + By) êy + (Az + Bz) êz.

The dot or scalar product between two vectors �A and �B is defined by
�A · �B=A B cos θ,

where A=
√

A2
x + A2

y + A2
z and B =

√
B2

x + B2
y + B2

z are the magnitudes of �A

and �B, respectively.
The cross or vector product of �A and �B, written as �A× �B, is another vector

whose magnitude | �A × �B| = A B sin θ. The direction of �A × �B is given by
the right-hand rule. Put the fingers of the right hand along �A and curl them
towards �B in the direction of the smaller angle between �A and �B. The thumb
then points in the direction of the new vector.

1The angle between the unit vectors is 90◦.

3.1.1 Bobby Blowfly

When you wanted to go Somewhere, And ended up going Nowhere,
The chances are strong, That you were heading for Erehwon.
An anonymous bard

So that he can better survey the goodies on the various picnic tables below
him, Bobby Blowfly is spiraling upwards along a trajectory described by the
Cartesian coordinates x=2 cos(t), y=sin(t), and z=2 t/3, where z (in meters)
is measured upwards and t is the time in seconds.

(a) Forming Bobby’s position vector �r, calculate his velocity �v, acceleration
�a, and speed V at time t.

(b) Determine �r, �v, �a, and V at t=3.1 and 17.3 s.

(c) Calculate the magnitude of his displacement for the interval 3.1 to 17.3 s.

(d) Calculate the distance he travels along the spiral path during the interval.

(e) Calculate the angle in radians and degrees between the velocity vectors
at the two times.

(f) Produce a 3-dimensional plot of his trajectory over the time interval with
his velocity and acceleration vectors indicated by arrows at the two times.

To solve this vector problem, the VectorCalculus library package is loaded.
When combined with the plots package, several warning messages will appear
on execution of the following command line which, recall, can be removed by
preceding the package commands with interface(warnlevel=0).

> restart: with(plots): with(VectorCalculus):

Bobby’s x, y, and z coordinates at time t are entered.
> x:=2*cos(t): y:=sin(t): z:=2*t/3:

Bobby’s position vector �r is entered, using the short-hand2 syntax <x,y,z>,
and his velocity �v and acceleration �a are calculated. Note that vector symbols
are not used in the Maple entries.

> r:=<x,y,z>; v:=diff(r,t); a:=diff(r,t,t);

r := 2 cos(t) ex + sin(t) ey +
2 t

3
ez

v := −2 sin(t) ex + cos(t) ey +
2
3

ez a := −2 cos(t) ex − sin(t) ey

Since no coordinate system has been specified, the default output is expressed in
terms of the Cartesian unit vectors ex, ey, and ez. On the computer screen, these
symbols are bold-faced. Using the DotProduct command in the VectorCalculus
package, Bobby’s speed V =

√
�v · �v is calculated. The “long form” of this

command is used here. A short-hand syntax will be introduced shortly.
> V:=sqrt(DotProduct(v,v));

2A longer form is Vector([x,y,z]).

V :=
1
3

√
4 + 36 sin(t)2 + 9 cos(t)2

The two times, T1 =3.1 s and T2 =17.3 s, are specified and an arrow operator
F formed for evaluating an arbitrary specified function f at a time t=T .

> T1:=3.1: T2:=17.3: F:=(f,T)->eval(f,t=T):

Making use of F, then �r, �v, �a, and V are determined at time T1 ,
> r1:=F(r,T1); v1:=F(v,T1); a1:=F(a,T1); V1:=F(V,T1);

r1 := (−1.998270301) ex + 0.04158066243 ey + 2.066666667 ez

v1 := (−0.08316132486) ex − 0.9991351503 ey +
2
3

ez

a1 := 1.998270301 ex − 0.04158066243 ey V1 := 1.204006353
and at time T2 (output suppressed here).

> r2:=F(r,T2); v2:=F(v,T2); a2:=F(a,T2); V2:=F(V,T2);

Bobby’s displacement �R =�r2 − �r1 in the time interval T2 − T1 is calculated
(output suppressed) along with its magnitude Rmag =

√
�R · �R. To mimic the

hand notation, the short-hand dot syntax is now used to enter the dot product.
Inserting spaces before and after the dot help to distinguish the dot product
from a decimal point and make for easier readability.

> R:=r2-r1: Rmag:=sqrt(R . R);

Rmag := 9.739961509
The magnitude of Bobby’s displacement in the time interval is about 9.7 meters.
The distance d that he travels along the path is obtained by calculating the
integral

∫ T2
T1 V dt,

> d:=int(V,t=T1..T2);

d := 23.93269942
and is found to be about 24 meters. This distance is considerably more than
the magnitude of the displacement.

From the definition of the dot product, the angle θ between the velocities �v1
and �v2 at times T1 and T2 is obtained by calculating arccos((�v1 ·�v2 )/(V1V2 )),
where V1 and V2 are the speeds.

> theta:=arccos((v1 . v2)/(V1*V2));

θ := 1.469381108
The angle between �v1 and �v2 is about 1.47 radians or, on converting from
radians to degrees, about 84 ◦.

> theta2:=convert(theta,units,radian,degree);

θ2 := 84.18933595
A 3-dimensional plot of Bobby’s trajectory over the time interval T1 to T2
is produced with the spacecurve command. To obtain a smooth curve, 500
plotting points are used. The shading=Z option is used to vary the color of the
trajectory with increasing vertical height z.

> gr||1:=spacecurve([x,y,z],t=T1..T2,numpoints=500,shading=Z):

A functional operator f is formed, involving the arrow command, to produce a
cylindrically shaped arrow representing the vector B with its tail at A. The color
c of the arrow must also be provided. The width of the arrow’s “body” as well
as its head width and head length are specified. These values are obtained by
trial and error on viewing the final figure.

> f:=(A,B,c)->arrow(A,B,shape=cylindrical_arrow,color=c,

width=0.1,head width=0.3,head length=0.5):
Then f is used to produce a green arrow for the velocity �v1 with its tail at �r1 ,
a green arrow for �v2 with its tail at �r2 , a red arrow for the acceleration �a1
with its tail at �r1 , and a red arrow for �a2 with its tail at �r2 .

> gr||2:=f(r1,v1,green): gr||3:=f(r2,v2,green):

gr||4:=f(r1,a1,red): gr||5:=f(r2,a2,red):
Using the sequence command, seq, the five graphs are put into a Maple set and
superimposed in Figure 3.1 with the display command.

> display({seq(gr||i,i=1..5)},axes=normal,labels=["x","y","z"]);

z

12

–1

1
y

–2

2
x

Figure 3.1: Bobby’s spiral path with �v and �a at T1 and T2 shown.

As expected, the velocity vectors are tangent to the trajectory, while the ac-
celeration vectors point to the center of the helix. The 3-dimensional plot may
be rotated on the computer screen by dragging with the mouse. It should be
noted that the plot has not been constrained.

3.1.2 Hiking in the Southern Chilkotin

How can you tell that “plop” on the trail is due to a grizzly bear?
It’s full of hiker’s “bear bells” and pepper spray.
Variation on an anonymous backpacker’s saying, R.I.P.

Over the years, I have been fortunate to hike and backpack throughout western
North America, in the mountains of Peru and southern Australia, and in the
jungles of Indonesia. Despite my encounters with a timber wolf, a giant grizzly

bear and many not-so-cuddly black bears, one of my favorite backpacking areas
is in the southern Chilkotin area of southwestern British Columbia. This region
marks the transition zone from the wet coastal rain forest with jagged glaciated
peaks to the west and the drier aspen covered slopes climbing towards smoother
reddish brown volcanic tops to the east. This recipe is inspired by my treks in
the southern Chilkotin.

After loading the plots and VectorCalculus packages, the height profile h of
a representative trekking region is entered. Positive x is to the east, positive y
to the north, with all distances in km.

> restart: with(plots): with(VectorCalculus):

> h:=(x̂ 2+ŷ 2)*exp(-0.5*(x̂ 2+ŷ 2))+1.2*exp(-(x-2.9)̂ 2-(y-2)̂ 2);

h := (x2 + y2) e(−0.5 x2−0.5 y2) + 1.2 e(−(x−2.9)2−(y−2)2)

To visualize the terrain, a 3-dimensional contour plot of h is produced over the
horizontal range x = −4 to 5 and y = −4 to 5, with 30 contours shown. The
default is 8 contours. If desired, the heights of the contours can be specified.
The option filled=true fills in the surface of the terrain. The color shading is
varied in the z direction. A particular angular orientation has been chosen for
the plot, but the computer picture may be rotated by dragging with the mouse.
The scaling has been left unconstrained to emphasize the vertical features.

> contourplot3d(h,x=-4..5,y=-4..5,contours=30,filled=true,

shading=z,axes=box,view=[-4..5,-4..5,0..1.3],
tickmarks=[2,3,3],orientation=[-100,60]);

Figure 3.2 shows a volcanic crater with an adjacent hill. Although the location

0 5
x–4

0

4

y

0

1

Figure 3.2: Contour plot of trekking region.

and height of the top of the hill can be approximately determined from the
picture, more precise values may be found by using the gradient operator, viz.,

gradh ≡ ∇ h =
∂h

∂x
êx +

∂h

∂y
êy.

At the top of the hill, the gradient is zero. The Gradient command is used to
calculate grad h at an arbitrary point (x, y). Alternately, one can obtain the
same result by using Del to calculate ∇ h.

> G:=Gradient(h,[x,y]); #alternately, G:=Del(h,[x,y]);

G := (2x e(−0.5 x2−0.5 y2) − 1.0 (x2 + y2) x e(−0.5 x2−0.5 y2)

+ 1.2 (−2 x + 5.8) e(−(x−2.9)2−(y−2)2)) ex + (2 y e(−0.5 x2−0.5 y2)

− 1.0 (x2 + y2) y e(−0.5 x2−0.5 y2) + 1.2 (−2 y + 4) e(−(x−2.9)2−(y−2)2)) ey

Notice that overbars appear above the unit (basis) vectors in the output of
the gradient operation. This indicates that G is a vector field defined at all
points (x, y). It is Maple’s way of reminding you that for general curvilinear
coordinate systems, such as the spherical polar system, the directions of the
unit vectors will vary from point to point in space. Only Cartesian unit vectors
are independent of position.

Guided by the figure, a numerical search is made over the range x=2 to 4
and y=1 to 4, using the floating point solve command, to find the zeros of the
x and y components (G[1] and G[2]) of G. The solution sol is assigned,

> sol:=fsolve({G[1],G[2]},{x,y},{x=2..4,y=1..4}); assign(sol):

sol := {y = 1.980898405, x = 2.872302687}
and the x and y coordinates of the peak of the hill are given by xmax and ymax .
The height of the hill, hmax , is obtained by evaluating h at xmax , ymax .

> xmax:=x; ymax:=y; hmax:=eval(h,{x=xmax,y=ymax});
xmax := 2.872302687 ymax := 1.980898405 hmax := 1.226303365

The elevation of the hill’s peak is about 1226 m. x and y are now unassigned.
> unassign(’x’,’y’):

Topographical maps used in hiking are 2-dimensional in nature. A 2-d contour
map with 25 contours is now made in cp with contourplot. The grid option is
set to 50 × 50, which generates 2500 equally spaced grid points for the contour
plot. The default is 25 × 25, which produces 625 grid points.

> cp:=contourplot(h,x=-4..5,y=-4..5,contours=25,grid=[50,50],

filled=true):
The fieldplot command is used in fp to plot the gradient G, placing thick
magenta colored arrows at equally spaced grid points (here 15 × 15), each arrow
pointing in the direction of increasing gradient at the grid point. The size of the
arrow is a measure of the strength of the gradient, larger arrows corresponding
to steeper gradients.

> fp:=fieldplot(G,x=-4..5,y=-4..5,arrows=THICK,grid=[15,15],

color=magenta):
The textplot command is used in tp to place the word “Top”, colored blue,
at the location xmax − 0.3, ymax + 0.1, adjacent to the top of the hill.

> tp:=textplot([[xmax-0.3,ymax+0.1,"Top"]],color=blue):

After stopping for lunch at the location (xmax − 2, ymax − 1.6) near the lip of
the crater, it is desired to hike to the top of the hill. The pointplot command
is used in pp to place size 16 green circles on the contour map at these locations.

> pp:=pointplot([[xmax,ymax],[xmax-2,ymax-1.6]],

symbol=circle,symbolsize=16,color=green):
The four graphs, fp, cp, tp, and pp, are superimposed with the display com-
mand, the resulting plot being shown in Figure 3.3.

> display([fp,cp,tp,pp],tickmarks=[3,5]);

Figure 3.3: Two-dimensional contour plot with gradient arrows.

As expected, the gradient arrows are perpendicular to the contour lines. The
gradient at the starting point xmax − 2, ymax − 1.6 is determined.

> G2:=eval(G,{x=xmax-2,y=ymax-1.6});
G2 := 0.6124645476 ex + 0.2695348694 ey

The slope at this point is obtained by calculating
√

�G2 · �G2 .
> Slope:=sqrt(G2 . G2);

Slope := 0.6691501086
The slope is about 0.67. It is positive, in agreement with the figure, indicating
that if we follow the gradient we will be initially climbing upwards from just
inside the lip of the crater. The slope can be expressed in radians by taking the
arctangent of the slope, which then can be converted to degrees.

> angle:=arctan(Slope);

angle := 0.5897199394
> angle:=convert(angle,units,radian,degree);

angle := 33.78846362

Initially we would be climbing at an angle of about 0.6 radians or 34◦ with
the horizontal. The initial direction of travel can be qualitatively deduced from
Figure 3.3. Quantitatively, the angle in radians measured with respect to the
x-axis (east) can be determined by calculating the arctangent of the ratio of
the y component of G2 to the x component.

> angle2:=evalf(arctan(G2[2]/G2[1]));

angle2 := 0.4145759092
The angle is about 0.41 radians or, on converting to degrees,

> angle2:=convert(angle2,units,radian,degree);

angle2 := 23.75344988
about 24 degrees with respect to the easterly direction.

3.1.3 Establishing These Identities is Easy

Trying to define yourself is like trying to bite your own teeth.
Alan Watts, American philosopher on identity, (1915–73)

Vector identities play an important role in many areas of mathematical physics,
particularly in electromagnetism. This recipe illustrates some examples involv-
ing dot and cross products in Cartesian coordinates.

Prove the following identities in Cartesian coordinates for arbitrary �A, �B, �C:

(a) �A · ( �B × �C) = ( �A × �B) · �C

(b) �A × ( �B × �C) = �B( �A · �C) − �C( �A · �B)

(c) [ �A × ( �B × �C)] + [ �B × (�C × �A)] + [�C × ( �A × �B)] = 0

(d) ( �B × �C) × (�C × �A) = �C( �A · ( �B × �C))
The VectorCalculus package is loaded and arbitrary vectors �A, �B, �C entered.

> restart: with(VectorCalculus):

> A:=<Ax,Ay,Az>; B:=<Bx,By,Bz>; C:=<Cx,Cy,Cz>;

A := Ax ex + Ay ey + Az ez

B := Bx ex + By ey + Bz ez

C := Cx ex + Cy ey + Cz ez

Using the long forms of the dot and cross product commands, the difference
between the left-hand and right-hand sides in (a) is calculated in id1 .

> id1:=DotProduct(A,CrossProduct(B,C))
-DotProduct(CrossProduct(A,B),C);

id1 := Ax (By Cz − Bz Cy) + Ay (Bz Cx − Bx Cz ) + Az (Bx Cy − By Cx )
− (Ay Bz − Az By)Cx − (Az Bx − Ax Bz )Cy − (Ax By − Ay Bx )Cz

Applying simplify to id1 yields zero, thus confirming the first identity.
> id1:=simplify(id1);

id1 := 0
For the remainder of the identities, the short-hand syntax for the dot and
crossproducts will be used. We have already used the dot notation for the dot
product. The short form for the cross product is &x. To prove the second
identity3 in (b), the left-hand side of the identity is now entered. Again, for
clarity, I have left spaces around the cross product symbols.

> LHS:= A &x (B &x C);

LHS := (Ay (Bx Cy − By Cx ) − Az (Bz Cx − Bx Cz )) ex

+ (Az (By Cz − Bz Cy) − Ax (Bx Cy − By Cx )) ey

+ (Ax (Bz Cx − Bx Cz ) − Ay (By Cz − Bz Cy)) ez
Similarly, the right-hand side is entered.

> RHS:=B*(A . C) - C*(A . B);

RHS := ((Ax Cx + Ay Cy + Az Cz )Bx − (Ax Bx + Ay By + Az Bz )Cx ) ex

+ ((Ax Cx + Ay Cy + Az Cz )By − (Ax Bx + Ay By + Az Bz )Cy) ey

+ ((Ax Cx + Ay Cy + Az Cz )Bz − (Ax Bx + Ay By + Az Bz )Cz ) ez

On simplifying LHS − RHS , a zero vector results in id2 , thus confirming (b).
> id2:=simplify(LHS - RHS);

id2 := 0 ex

The remaining two identities in (c) and (d) are proven in a similar manner as
shown in id3 and id4 , respectively. For the latter, the difference between the
left-hand and right-hand sides is calculated.

> id3:=(A &x (B &x C)) + (B &x (C &x A)) + (C &x (A &x B));

> id3:=simplify(id3);

id3 := 0 ex

> id4:=(B &x C) &x (C &x A) - C*(A . (B &x C));

> id4:=simplify(id4);

id4 := 0 ex

3.1.4 This Task is Not a Chore

The hardest task of a girl’s life, nowadays, is to prove
to a man that his intentions are serious.
Helen Rowland, American journalist, A Guide to Men,Intermezzo (1922)

In classical mechanics, a standard task is to express the velocity and accel-
eration of a particle in some other curvilinear coordinate system, in particular
in spherical polar and cylindrical coordinates. Curvilinear coordinate systems
will be dealt with at length in the next section, but let me show you how easy

3Often referred to as the BAC–CAB rule, because of the structure of the right-hand side,

it is to obtain the velocity and acceleration of a particle in spherical polar co-
ordinates. The relation between the Cartesian coordinates (x, y, z) and the
spherical coordinates (r, θ, φ) is given by x=r sin θ cos φ, y=r sin θ sin φ, and
z=r cos θ, where r is the radial distance from the origin, θ is the angle between
the radius vector and the z-axis, and φ is the angle that the projection of the
radius vector into the x-y plane makes with the x-axis. The ranges of these
coordinates are 0 ≤ r < ∞, 0 ≤ θ ≤ π, and 0 ≤ φ ≤ 2π.

After loading the VectorCalculus package,
> restart: with(VectorCalculus):

the relations between the two coordinates systems are entered, the spherical
variables taken to be time-dependent so that time derivatives can be taken.

> X:=r(t)*sin(theta(t))*cos(phi(t)):

Y:=r(t)*sin(theta(t))*sin(phi(t)): Z:=r(t)*cos(theta(t)):
The position vector �R=X êx+Y êy+Z êz of a particle is entered as a vector field
in Cartesian coordinates, the above relations being automatically substituted.

> R:=VectorField(<X,Y,Z>,’cartesian’[x,y,z]);

R := r(t) sin(θ(t)) cos(φ(t)) ex + r(t) sin(θ(t)) sin(φ(t)) ey + r(t) cos(θ(t)) ez

The velocity �v and acceleration �a of the particle are calculated by differentiating
�R once and twice, respectively, with respect to t. A line-ending colon has been
placed on the acceleration to suppress the very lengthy output. This will be
done for all intermediate steps involving the acceleration.

> v:=diff(R,t); a:=diff(R,t,t):

v := ((
d

dt
r(t)) sin(θ(t)) cos(φ(t)) + r(t) cos(θ(t)) (

d

dt
θ(t)) cos(φ(t))

− r(t) sin(θ(t)) sin(φ(t)) (
d

dt
φ(t))) ex

+((
d

dt
r(t)) sin(θ(t)) sin(φ(t)) + r(t) cos(θ(t)) (

d

dt
θ(t)) sin(φ(t))

+ r(t) sin(θ(t)) cos(φ(t)) (
d

dt
φ(t))) ey

+((
d

dt
r(t)) cos(θ(t)) − r(t) sin(θ(t)) (

d

dt
θ(t))) ez

The MapToBasis command can be used to express an arbitrary Cartesian unit
vector û in terms of the spherical (polar) unit vectors êr, êθ, êφ. A functional
operator F is formed to do this.

> F:=u->MapToBasis(u,’spherical’[r,theta,phi]):

The operator F is applied to the velocity and acceleration. The symbols %1
and %2 which appear in v2 indicate sub-expressions. This form of the output
is an artifact of exporting Maple into the text as Latex output. Latex is the
standard word processing language used in preparing scientific documents, such
as this text, which involve mathematical expressions.

> v2:=F(v); a2:=F(a):

v2 := (%2 sin(θ) cos(φ) + %1 sin(φ) sin(θ)

+ ((
d

dt
r(t)) cos(θ(t)) − r(t) sin(θ(t)) (

d

dt
θ(t))) cos(θ)) er

+(%2 cos(φ) cos(θ) + %1 sin(φ) cos(θ)

− ((
d

dt
r(t)) cos(θ(t)) − r(t) sin(θ(t)) (

d

dt
θ(t))) sin(θ)) eθ

+(−%2 sin(φ) + %1 cos(φ)) eφ

%1 := (
d

dt
r(t)) sin(θ(t)) sin(φ(t)) + r(t) cos(θ(t)) (

d

dt
θ(t)) sin(φ(t))

+ r(t) sin(θ(t)) cos(φ(t)) (
d

dt
φ(t))

%2 := (
d

dt
r(t)) sin(θ(t)) cos(φ(t)) + r(t) cos(θ(t)) (

d

dt
θ(t)) cos(φ(t))

− r(t) sin(θ(t)) sin(φ(t)) (
d

dt
φ(t))

The MapToBasis command has generated trigonometric terms in v2 (and a2 )
which are not time-dependent. To simplify the velocity and acceleration, an
arrow operator G is created to substitute the requisite time-dependence.

> G:=u->subs({sin(theta)=sin(theta(t)),cos(theta)=
cos(theta(t)),cos(phi)=cos(phi(t)),sin(phi)=sin(phi(t))},u):

Then G is applied to v2 and a2 and the results simplified with the trig option.

> v3:=simplify(G(v2),trig); a3:=simplify(G(a2),trig):

v3 := (
d

dt
r(t)) er + r(t) (

d

dt
θ(t)) eθ + r(t) sin(θ(t)) (

d

dt
φ(t)) eφ

The velocity expression given in v3 is the standard result [FC99] in spherical
polar coordinates. The standard result for the acceleration given in a4 follows
on making the algebraic substitution cos2 θ(t)=1 − sin2 θ(t).

> a4:=algsubs(cos(theta(t))ˆ2=1-sin(theta(t))ˆ2,a3);

a4 := ((
d2

dt2
r(t)) − r(t) (

d

dt
φ(t))2 sin(θ(t))2 − r(t) (

d

dt
θ(t))2) er

+(r(t) (
d2

dt2
θ(t)) + 2 (

d

dt
r(t)) (

d

dt
θ(t)) − r(t) sin(θ(t)) cos(θ(t)) (

d

dt
φ(t))2) eθ

+ (r(t) sin(θ(t)) (
d2

dt2
φ(t)) + 2 (

d

dt
r(t)) sin(θ(t)) (

d

dt
φ(t))

+ 2 r(t) cos(θ(t)) (
d

dt
θ(t)) (

d

dt
φ(t))) eφ

The above results may also be obtained by making use of the LinearAlgebra
package, as illustrated in Supplementary Recipe 03-S04. The velocity and
acceleration are just as easily derived in cylindrical coordinates.

3.2 Vectors: Curvilinear Coordinates

Consider a general 3-dimensional orthogonal coordinate system with coordi-
nates u = u(x, y, z), v = v(x, y, z), w = w(x, y, z). For example, for spherical
(polar) coordinates, u is the radial distance r, v is the polar angle θ, and w is
the azimuthal angle φ. The differential element of length in, say, the u direction
is given by dsu =hu du, where the scale factor hu is given by

hu =

√(
∂x

∂u

)2

+
(

∂y

∂u

)2

+
(

∂z

∂u

)2

with similar expressions for hv and hw.
The unit vectors êu, êv, êw are related to the Cartesian unit vectors by

êu =
∂�r

∂su
=

1
hu

∂�r

∂u
, êv =

1
hv

∂�r

∂v
, êw =

1
hw

∂�r

∂w
,

where �r=x êx + y êy + z êz is the position vector with x=x(u, v, w), etc.
The element of area on, say, a surface of constant u is dAu = dsv dsw =

huhvdu dv while the volume element is dV =dsudsvdsw =huhvhwdu dv dw.
The gradient (∇), divergence (∇·), curl (∇×), and Laplacian (∇ · ∇ ≡ ∇2)

operators are calculated as follows [Gri99]:

∇f =
êu

hu

∂f

∂u
+

êv

hv

∂f

∂v
+

êw

hw

∂f

∂w
,

∇· �A = [
∂

∂u
(Au hv hw) +

∂

∂v
(Av hu hw) +

∂

∂w
(Aw hu hv) ]/(hu hv hw),

∇× �A = [huêu[
∂

∂v
(hwAw) − ∂

∂w
(hvAv) ] + hv êv[

∂

∂w
(huAu) − ∂

∂u
(hwAw) ]

+ hw êw[
∂

∂u
(hvAv) − ∂

∂v
(huAu) ] ]/(hu hv hw),

∇2 f = [
∂

∂u
(
hvhw

hu

∂f

∂u
) +

∂

∂v
(
huhw

hv

∂f

∂v
) +

∂

∂w
(
huhv

hw

∂f

∂w
) ]/(hu hv hw).

3.2.1 From Scale Factors to Vector Operators

Everything that is beautiful and noble is the product
of reason and calculation.
Charles Baudelaire, French poet, (1821–67)

In this recipe, functional operators are created to calculate the scale factors
and the vector operators for any 3-dimensional orthogonal coordinate system.
As an application, the volume of a sphere, and the gradient, divergence, curl,
and Laplacian are calculated for spherical polar coordinates (r, θ, φ).

To make the entries easier, the aliases t and p are used for theta and phi.
> restart: alias(theta=t,phi=p):

The arrow operator H is introduced to calculate the scale factor and simplify it
for a given input coordinate u.

> H:=u->simplify(sqrt(diff(x,u)ˆ2+diff(y,u)ˆ2+diff(z,u)ˆ2),

symbolic):
The relations between the Cartesian coordinates (x, y, z) and (r, θ, φ) are
entered. Then the spherical polar scale factors hr, hθ, and hφ are calculated.

> x:=r*cos(p)*sin(t); y:=r*sin(p)*sin(t); z:=r*cos(t);

x := r cos(φ) sin(θ) y := r sin(φ) sin(θ) z := r cos(θ)
> h[r]:=H(r); h[t]:=H(t); h[p]:=H(p);

hr := 1 hθ := r hφ := r sin(θ)
With all the scale factors now determined, the volume of a sphere of radius R

is given by V =
∫ 2π

0

∫ π

0

∫ R

0 hr hθ hφ dr dθ dφ. This integral is now entered,
> V:=Int(Int(Int(h[r]*h[t]*h[p],r=0..R),t=0..Pi),p=0..2*Pi);

V :=
∫ 2 π

0

∫ π

0

∫ R

0
r2 sin(θ) dr dθ dφ

and evaluated with the value command.
> Volume:=value(V);

Volume :=
4 R3 π

3
The well-known formula for the volume of a sphere results.

A functional operator g is formed for calculating the gradient of a general
function f(u, v, w) for arbitrary input coordinates u, v, and w.

> g:=(u,v,w)->e[u]*diff(f(u,v,w),u)/h[u]

+e[v]*diff(f(u,v,w),v)/h[v]+e[w]*diff(f(u,v,w),w)/h[w];

g :=(u, v, w) → eu diff(f(u, v, w), u)
hu

+
ev diff(f(u, v, w), v)

hv
+

ew diff(f(u, v, w), w)
hw

Similarly, arrow operators d and c are created for calculating the diver-
gence and curl of a general vector field �A=Au(u, v, w) êu + Av(u, v, w) êv +
Aw(u, v, w) êw for arbitrary coordinates u, v, and w.

> d:=(u,v,w)->(diff(A[u](u,v,w)*h[v]*h[w],u)
+ diff(A[v](u,v,w)*h[u]*h[w],v)
+ diff(A[w](u,v,w)*h[u]*h[v],w))/(h[u]*h[v]*h[w]):

> c:=(u,v,w)-> (1/(h[u]*h[v]*h[w]))*
(h[u]*e[u]*(diff(h[w]*A[w](u,v,w),v)-diff(h[v]*A[v](u,v,w),w))
+h[v]*e[v]*(diff(h[u]*A[u](u,v,w),w)-diff(h[w]*A[w](u,v,w),u))
+h[w]*e[w]*(diff(h[v]*A[v](u,v,w),u)-diff(h[u]*A[u](u,v,w),v))):

Finally, here’s an operator L for calculating the Laplacian of f(u, v, w).

> L:=(u,v,w)->(diff(h[v]*h[w]*diff(f(u,v,w),u)/h[u],u)
+diff(h[u]*h[w]*diff(f(u,v,w),v)/h[v],v)
+diff(h[u]*h[v]*diff(f(u,v,w),w)/h[w],w))/(h[u]*h[v]*h[w]):

The forms of the gradient and Laplacian are calculated for spherical polar
coordinates, the divergence and curl being left for you to do as an exercise.

> gradf:=g(r,t,p); Lapf:=expand(L(r,t,p));

gradf := er (
∂

∂r
f(r, θ, φ)) +

eθ (
∂

∂θ
f(r, θ, φ))

r
+

eφ (
∂

∂φ
f(r, θ, φ))

r sin(θ)

Lapf :=
2 (

∂

∂r
f(r, θ, φ))

r
+ (

∂2

∂r2 f(r, θ, φ)) +
cos(θ) (

∂

∂θ
f(r, θ, φ))

r2 sin(θ)

+

∂2

∂θ2 f(r, θ, φ)

r2 +

∂2

∂φ2 f(r, θ, φ)

r2 sin(θ)2
The above expressions agree with those found in standard mathematics texts.

3.2.2 Vector Operators the Easy Way

Civilization is not by any means an easy thing to attain to. There are
only two ways by which man can reach it. One is by being cultured,
the other by being corrupt.
Oscar Wilde, Anglo-Irish author, The Picture of Dorian Gray, 1891

I hope that I am not corrupting you by now demonstrating an easy way of
obtaining the vector operators, even for coordinate systems which initially may
be unfamiliar to you. It’s important that you know how the vector operators
are calculated from first principles, so I trust that you did not skip the last
recipe. But if you want to solve PDE boundary and initial value problems,
the main subject of the Entrees, it is desirable, e.g., to obtain the Laplacian
operator as quickly as possible.

In this recipe, I will show you how to gain information about any two- or
three-dimensional orthogonal coordinate system and illustrate how the gradient,
divergence, curl, and Laplacian are easily obtained for paraboloidal coordinates.

To calculate the fore-mentioned vector operators, the VectorCalculus pack-
age must first be loaded.

> restart: with(plots): with(VectorCalculus):

Executing the following command line will open a help page about the two-
and three-dimensional coordinate systems known to Maple.

> ?coords;

The majority of coordinate systems that you are likely to encounter in math-
ematical physics are listed on this help page. The relationship between the
given curvilinear coordinate system and Cartesian coordinates is stated. For
example, for the paraboloidal coordinates (u,v,w), one has

x = u v cos(w), y = u v sin(w), z = (u2 − v2)/2
with 0 ≤ u < ∞, 0 ≤ v < ∞, and 0 ≤ w ≤ 2 π.



3.2. VECTORS: CURVILINEAR COORDINATES 101

A common way of visualizing a coordinate system is to plot the surface
corresponding to holding each of the coordinates constant. For example, for
spherical polar coordinates, holding the radial coordinate r fixed generates a
spherical surface. Although we could create a recipe to do this task for us for
paraboloidal coordinates, an easier way is to use the command coordplot3d
which produces a representative surface corresponding to holding u, v, and w
fixed. Executing the following command line results in Figure 3.4.

> coordplot3d(paraboloidal,orientation=[-20,40],axes=frame,

tickmarks=[2,2,2],labels=["x","y","z"]);

–1

1

x

–1

1

y

z

Figure 3.4: Surfaces for fixed u, v, and w in paraboloidal coordinates.

The geometric structure of the three surfaces is easily understood. Let’s form
x2 + y2 = u2 v2 (cos2 w + sin2 w) = u2 v2, Then, holding, e.g., u fixed, we have
z = (u2 − v2)/2 = (u2 − (x2 + y2)/u2)/2 which, for fixed u, is the equation
of a paraboloid (parabola of revolution about the z-axis). Holding u and v
constant produces the two intersecting paraboloids shown in the picture. As
you may confirm, the half-plane corresponds to holding w fixed. For a thorough
discussion of curvilinear coordinate systems, Morse and Feshbach’s Methods of
Theoretical Physics ([MF53]) is highly recommended.

Entering the assumption that u > 0 and v > 0 to simplify the results,
> assume(u>0,v>0):

the coordinates u, v, and w are now set to be paraboloidal.
> SetCoordinates(’paraboloidal’[u,v,w]):

Using the Gradient operator, ∇f(u, v, w) is calculated and simplified.
> gradf:=simplify(Gradient(f(u,v,w)));

gradf :=

∂

∂u
f(u, v, w)

√
u2 + v2

eu +

∂

∂v
f(u, v, w)

√
u2 + v2

ev +

∂

∂w
f(u, v, w)

u v
ew

From the structure of the output, one can easily deduce that the scale factors
for paraboloidal coordinates are hu = hv =

√
u2 + v2 and hw = u v. Next,

∇2f(u, v, w) is calculated, simplified, and expanded.
> Lapf:=expand(simplify(Laplacian(f(u,v,w))));

Lapf :=

∂

∂u
f(u, v, w)

(u2 + v2) u
+

∂2

∂u2 f(u, v, w)

u2 + v2 +

∂

∂v
f(u, v, w)

(u2 + v2) v
+

∂2

∂v2 f(u, v, w)

u2 + v2

+

∂2

∂w2 f(u, v, w)

(u2 + v2) v2 +

∂2

∂w2 f(u, v, w)

(u2 + v2) u2

Entering a general vector field �A in terms of the coordinates u, v, w,
> A:=VectorField(<Au(u,v,w),Av(u,v,w),Aw(u,v,w)>);

A := Au(u, v, w) eu + Av(u, v, w) ev + Aw(u, v, w) ew

div �A and curl �A are calculated using the Divergence and Curl operators.
> divA:=expand(simplify(Divergence(A)));

> curlA:=expand(simplify(Curl(A)));

The lengthy outputs have been suppressed here, so you will have to execute the
above command line on the computer to see the results.

3.2.3 These Operators Do Not Have an Identity Crisis

It is always the same: once you are liberated,
you are forced to ask who you are.
Jean Baudrillard, French semiologist, America,“Astral America” (1986)

Now that you have learned how to generate vector operators, you should never
experience a mathematical crisis in proving a vector operator identity in any
orthogonal coordinate system. This recipe provides some examples.

Prove the following identities in the indicated coordinate systems:

(a) ∇ · (∇ × �A) = 0, Spherical polar ( �A = �A(r, θ, φ))

(b) ∇ × (∇f) = 0, Cylindrical (f = f(ρ, φ, z))

(c) ∇ × (∇ × �A) = ∇(∇ · �A) − ∇2 �A, Paraboloidal ( �A = �A(u, v, w))

All three identities involve non-Cartesian coordinate systems, so the coordinates
must be specified in each case. For (a), the spherical polar coordinates r, θ, φ
are required, which are entered and assigned the name sp. The coordinates are
then set to be spherical.

> restart: with(VectorCalculus):

> sp:=r,theta,phi: SetCoordinates(’spherical’[sp]):

A general vector field �A is entered in spherical coordinates with components
labeled Ar, At, and Ap.

> A:=VectorField(<Ar(sp),At(sp),Ap(sp)>);

A := Ar(r, θ, φ) er + At(r, θ, φ) eθ + Ap(r, θ, φ) eφ

The left-hand side of (a) is just div(curl �A). Entering Divergence(Curl(A))
yields zero, confirming the first vector identity.

> id1:=Divergence(Curl(A));

id1 := 0
For (b), the coordinates ρ, φ, z are set to be cylindrical.

> SetCoordinates(’cylindrical’[rho,phi,z]):

The left-hand side of the identity in (b) is just curl gradf , so one could enter
Curl(Gradient(f(rho,phi,z)). An alternate way is to note that the Del
operator can be used for calculating ∇. Then using the short-hand syntax
for the cross product, we can enter Del &x Del(f(rho,phi,z)). The result of
executing the command line is zero, confirming the second identity.

> id2:=Del &x Del(f(rho,phi,z));

id2 := 0 eρ

The identity in (c) involves paraboloidal coordinates u, v, w which are set
> SetCoordinates(’paraboloidal’[u,v,w]):

and a general vector field �A2 entered.
> A2:=VectorField(<Au(u,v,w),Av(u,v,w),Aw(u,v,w)>);

A2 := Au(u, v, w) eu + Av(u, v, w) ev + Aw(u, v, w) ew

Using Del and the short-hand syntax for both the dot and cross products, the
right-hand side is subtracted from the left-hand side in (c).

> id3:=Del &x (Del &x A2)-Del(Del . A2)+(Del . Del)(A2);

id3 := 0 eu

The result is zero, confirming the third identity.

3.2.4 Is This Vector Field Conservative?

The most radical revolutionary will become a conservative the day
after the revolution.
Hannah Arendt, American political philosopher, (1906–1975)

Conservative vector fields play an important role in physics. A conservative
field �A is characterized by having a closed line integral

∮
�A · d�� = 0 around an

arbitrary path, or from Stokes’s theorem, ∇× �A=0 everywhere. See Recipe 03-
S10. From the identity ∇×∇φ=0 for any function φ, one can write �A=−∇φ,
where φ is the potential and the minus sign is inserted by convention. Given
a conservative �A(�r), φ(�r) may be obtained to within an arbitrary constant by
performing the line integral from an arbitrary reference point to �r.

Consider the following vector field,
�A = (2x sin y + 2 y2 − 12 x3y3 + 2 x y z − 3 z) êx

+ ((1 + x2)(z + cos y) − 3 y2 + 4 x y − 9 x4y2) êy + ((1 + x2) y − 3 x) êz.

Show that �A is conservative and determine the corresponding potential φ(x, y, z).
Determine the potential φ1 which passes through x=0.13, y=1.82, z =−1.24.
Rounding φ1 to the nearest integer Φ, plot the equipotentials corresponding to
±Φ and superimpose the graph on a plot of the vector field.

Here is the solution provided by Ms. I. M. Curious. She loads the plots and
VectorCalculus packages and sets the coordinates to be Cartesian.

> restart: with(plots): with(VectorCalculus):

> SetCoordinates(’cartesian’[x,y,z]):

The given vector field �A is entered.
> A:= VectorField(<2*x*sin(y)+2*yˆ2-12*xˆ3*yˆ3+2*x*y*z-3*z,

(1+xˆ2)*(z+cos(y))-3*yˆ2+4*x*y-9*xˆ4*yˆ2,(1+xˆ2)*y-3*x>);

A := (2x sin(y) + 2 y2 − 12 x3 y3 + 2 x y z − 3 z) ex+
((x2 + 1) (z + cos(y)) − 3 y2 + 4 x y − 9 x4 y2) ey + ((x2 + 1) y − 3 x) ez

Taking the curl of �A and simplifying, yields a zero vector in C, confirming that
�A is a conservative field.

> C:=simplify(Curl(A));

C := 0 ex

Choosing (0,0,0) as the reference point, φ(x, y, z) is obtained by using the line
integral command to integrate �A along a straight line from (0,0,0) to (x, y, z).

> phi:=-LineInt(A,Line(<0,0,0>,<x,y,z>));

φ := 3x z − y x2 z + 3 x4 y3 − y z − 2 x y2 − x2 sin(y) + y3 − sin(y)
The potential is evaluated at x=0.13, y=1.82, z=−1.24,

> phi1:=eval(phi,{x=0.13,y=1.82,z=-1.24});
φ1 := 5.998362305

and rounded off to the nearest integer.
> Phi:=round(phi1);

Φ := 6
The implicitplot3d command is used in gr1 to plot the equipotential surfaces
corresponding to φ=Φ=6 and φ=−Φ. The grid is taken to be 20 × 20 × 20,
a patchcontour style is used, and the shading is zhue.

> gr1:=implicitplot3d({phi=Phi,phi=-Phi},x=-2.5..2.5,
y=-2.5..2.5,z=-2.5..2.5,grid=[20,20,20],
style=patchcontour,shading=zhue):

The fieldplot3d command is used in gr2 to plot the vector field �A, the field
being represented by thick red arrows on a grid 7 × 7 × 7.

> gr2:=fieldplot3d(A,x=-2.5..2.5,y=-2.5..2.5,z=-2.5..2.5,

grid=[7,7,7],arrows=THICK,color=red):
I. M. then superimposes the two graphs with the display command, choosing
a particular orientation for the 3-dimensional viewing box.

> display({gr1,gr2},axes=boxed,orientation=[40,40],
tickmarks=[3,3,3],labels=["x","y","z"]);

Figure 3.5: Equipotential surfaces and vector field arrows.

The resulting black and white version of the computer plot is shown in Fig-
ure 3.5. The equipotential surfaces are quite convoluted in appearance, but it
appears that the vector field arrows are perpendicular to them as expected. I.
M. suggests that you experiment with the recipe, trying other equipotentials,
densities of arrows, etc. You could even change the form of the vector field.

3.2.5 The Divergence Theorem

Nothing leads the scientist so astray as a premature truth.
Jean Rostand, French biologist, Penses d’un Biologiste, 1939

The divergence (Gauss’s) theorem states that for a closed volume V having
a bounding surface S,

∫
V

(∇ · �A) dv =
∮

S
�A · d�a, where d�a is an element of area

on S, dv is an element of volume in V , and
∮

S
denotes a closed surface integral.

Consider the flow of a fluid characterized by the velocity vector field
�A = r cos θ êr + r sin θ êθ + r sin θ cos φ êφ.

in spherical coordinates. Is this a conservative field? Create a 3-dimensional
plot of the vector field. Check the divergence theorem for �A, using as your
volume V an inverted closed hemispherical bowl of radius R, resting on the x-y
plane and centered on the origin.

The plots and VectorCalculus packages are loaded and the coordinates r, θ,
and φ are set to be spherical. The given vector field �A is then entered.

> restart: with(plots): with(VectorCalculus):

> SetCoordinates(’spherical’[r,theta,phi]):

> A:=VectorField(<r*cos(theta),r*sin(theta),

r*sin(theta)*cos(phi)>);

A := r cos(θ) er + r sin(θ) eθ + r sin(θ) cos(φ) eφ

To plot the vector field, I will temporarily switch to Cartesian coordinates. This
can be accomplished with the following MapToBasis command.

> A2:=MapToBasis(A,’cartesian’[x,y,z]);

A2 := (
2 z x√

x2 + y2 + z2
− x y√

x2 + y2
) ex + (

2 z y√
x2 + y2 + z2

+
x2√

x2 + y2
) ey

+ (
z2√

x2 + y2 + z2
+

−x2 − y2√
x2 + y2 + z2

) ez

Then the fieldplot3d command is used to plot A2 , producing SLIM red ar-
rows on a 7 × 7 × 7 grid.

> fieldplot3d(A2,x=-2.5..2.5,y=-2.5..2.5,z=-2.5..2.5,

grid=[7,7,7],arrows=SLIM,color=red,axes=box);

Figure 3.6: Arrows representing the 3-dimensional vector field �A.

The resulting 3-dimensional picture is reproduced in Figure 3.6, but is best
viewed on the computer screen where it can be rotated by dragging with the
mouse. Looking at the picture, do you think that the vector field is conservative,
i.e., has zero curl? Let’s calculate the curl as well as the divergence which we
will need for confirming the divergence theorem.

> curl:=Del &x A; div:=simplify(Del . A);

curl := 2 cos(φ) cos(θ) er − 2 sin(θ) cos(φ) eθ + 3 sin(θ) eφ

div := 5 cos(θ) − sin(φ)
�A has a non-zero curl, and therefore is not conservative. Now, let’s check the
divergence theorem. The left-hand side

∫ 2π

0

∫ π/2
0

∫ R

0 (∇ · �A) dv of the divergence
theorem is entered for the volume contained in the hemispherical bowl

> LHS:=Int(Int(Int(div*rˆ2*sin(theta),r=0..R),

theta=0..Pi/2),phi=0..2*Pi);

LHS :=
∫ 2 π

0

∫ π
2

0

∫ R

0
r2 sin(θ) (5 cos(θ) − sin(φ)) dr dθ dφ

and then evaluated with the value command.
> LHS:=value(LHS);

LHS :=
5 R3 π

3
On the hemispherical upper surface of the bowl, the unit vector is radial every-
where, i.e., is given by êr. This unit vector is entered in n1 as a vector field. On
the planar bottom surface of the bowl, the unit vector points in the direction
êθ. This unit vector is entered in n2, also as a vector field.

> n1:=VectorField(<1,0,0>); n2:=VectorField(<0,1,0>);

n1 := er n2 := eθ

The vector field �A is evaluated on the top (hemisphere) surface in At and on
the bottom (planar) surface in Ab.

> At:=eval(A,r=R); Ab:=eval(A,theta=Pi/2);

At := R cos(θ) er + R sin(θ) eθ + R sin(θ) cos(φ) eφ

Ab := r eθ + r cos(φ) eφ

Now, we can evaluate the surface integral on the right-hand side of the diver-
gence theorem. The contribution from the hemispherical top is of the form∫ 2π

0

∫ π/2
0 (�At · n̂1 ) R2 sin θ dθ dφ, which is now entered and then evaluated.

> RHS1:=Int(Int((At . n1)*Rˆ2*sin(theta),theta=0..Pi/2),

phi=0..2*Pi);

RHS1 :=
∫ 2 π

0

∫ π
2

0
R3 cos(θ) sin(θ) dθ dφ

> RHS1:=value(RHS1);

RHS1 := R3 π

The contribution to the surface integral from the planar bottom is given by∫ 2π

0

∫ R

0 (�Ab · n̂2 ) r sin(π/2) dr dφ which is entered and evaluated.
> RHS2:=Int(Int((Ab . n2)*r*sin(Pi/2),r=0..R),phi=0..2*Pi);

RHS2 :=
∫ 2 π

0

∫ R

0
r2 dr dφ

> RHS2:=value(RHS2);

RHS2 :=
2 R3 π

3
Adding the two surface integral contributions, RHS1 and RHS2 , completes the
evaluation of the right-hand side of the divergence theorem.

> RHS:=RHS1+RHS2;

RHS :=
5 R3 π

3
The results RHS and LHS are identical, confirming the divergence theorem.

3.3 Matrices

A general matrix A of order m × n is of the form

A =

⎛
⎜⎜⎝

a11 a12 a13 ... a1n

a21 a22 a23 ... a2n

... ... ... ... ...
am1 am2 am3 ... amn

⎞
⎟⎟⎠

the index m labeling the row and n the column. Each number ajk in A is called
an element. If the number of columns equals the number of rows, the matrix
is said to be square. A matrix having only one row is called a row matrix or a
row vector, while a matrix having only one column is called a column matrix or
column vector. Some basic properties of matrices are as follows:

(a) Addition and Subtraction: If two matrices A=(ajk) and B =(bjk) are of
the same order, then A ± B=(ajk ± bjk).

(b) Multiplication: If A=(ajk) and λ is any scalar, then λA=A λ=(λajk).
If A=(ajk) is an m × n matrix and B=(bjk) is an n × p matrix, then the
product A B (or A · B) is a matrix C = (cjk), where cjk =

∑n
i=1 aji bik.

The new matrix C is of order m × p.

(c) Transpose: The transpose AT of a matrix A = (ajk) is AT = (akj), i.e.,
interchange rows and columns.

(d) Hermitian matrix: If A = (ajk), then the complex conjugate matrix is
A∗ =(a∗

jk). The Hermitian conjugate (or adjoint) matrix A† = (AT )∗. A
square matrix is Hermitian if A† = A.

(e) Inverse of a matrix: If A is a non-singular matrix (i.e., its determi-
nant det(A) �= 0), then the inverse matrix A−1 is of the form A−1 =
(Ajk)T /det(A), where (Ajk) is the matrix of cofactors Ajk. (The cofactor
Ajk is equal to (−1)j+k times the resulting determinant of A obtained by
removing all the elements of the jth row and kth column.) It follows that
A A−1 = A−1 A = I, where I is the unit matrix with each element along
its principal diagonal equal to 1 and all off-diagonal elements 0.

Other matrix properties will be introduced in the recipes of this section.

3.3.1 Some Matrix Basics

The basic tool for the manipulation of reality is the manipulation of
words. If you can control the meaning of words, you can control the
people who must use the words.
Philip K. Dick, American science fiction writer, (1928–82)

In this recipe, the basic Maple syntax for dealing with matrices is introduced.

If A=

⎛
⎝ 2 1 −1

1 −2 3
−2 1 2

⎞
⎠ and B=

⎛
⎝ 1 −1 2

−2 1 3
2 −1 1

⎞
⎠ show that

(a) A and B are non-singular matrices

(b) (A + B)2 = A2 + A B + B A + B2

(c) (A B)−1 = B−1 A−1

(d) (A B)T = BT AT

The basic Maple library package for dealing with matrices and determinants is
the LinearAlgebra package. Replace the colon with a semicolon in with(Linear
Algebra) if you wish to see the command structures available in this package.

> restart: with(LinearAlgebra):

A matrix can be entered in different ways. For example, A is now entered using
the Matrix command with the argument consisting of the elements arranged in
order in a list of lists.

> A:=Matrix([[2,1,-1],[1,-2,3],[-2,1,2]]);

A :=

⎡
⎣ 2 1 −1

1 −2 3
−2 1 2

⎤
⎦

The matrix is clearly a 3 × 3 (square) matrix. The dimensionality of A can be
confirmed using the Dimension command.

> Dimension(A);

3, 3
A matrix such as B can also be entered by using the following syntax.

> B:=<<1|-1|2>,<-2|1|3>,<2|-1|1>>;

B :=

⎡
⎣ 1 −1 2

−2 1 3
2 −1 1

⎤
⎦

(a) To prove that A and B are non-singular matrices, their determinants are
calculated and seen to be non-zero.

> Determinant(A); Determinant(B);

−19 − 4
(b) On the left-hand side, (A + B)2 is calculated.

> LHS:=(A+B)ˆ2;

LHS :=

⎡
⎣ 9 0 6

−2 1 11
0 0 9

⎤
⎦

To multiply two matrices A and B, one can use either the “long-hand” syntax
Multiply(A,B) or the shorter dot notation A . B, again leaving spaces either
side of the dot as we did for vector multiplication. The right-hand side A2 +
A B + B A + B2 is now entered using the “short-hand” syntax. For clarity, I
have place round brackets around each matrix multiplication.

> RHS:=Aˆ2+(A . B)+(B . A)+Bˆ2;

RHS :=

⎡
⎣ 9 0 6

−2 1 11
0 0 9

⎤
⎦

The matrix result in RHS is the same as in LHS , thus proving the relation.

(c) The inverse matrices A−1 and B−1 are determined.

> Ainverse:=MatrixInverse(A); Binverse:=MatrixInverse(B);

Ainverse :=

⎡
⎢⎢⎢⎢⎢⎣

7
19

3
19

−1
19

8
19

−2
19

7
19

3
19

4
19

5
19

⎤
⎥⎥⎥⎥⎥⎦ Binverse :=

⎡
⎢⎢⎢⎢⎢⎣

−1
1
4

5
4

−2
3
4

7
4

0
1
4

1
4

⎤
⎥⎥⎥⎥⎥⎦

Then (A B)−1 − B−1 A−1 is calculated in eq1 .

> eq1:=MatrixInverse(A . B)-(Binverse . Ainverse);

eq1 :=

⎡
⎣ 0 0 0

0 0 0
0 0 0

⎤
⎦

The result is a zero or null matrix, thus confirming the relation.

(d) The transposes AT and BT of A and B are determined.

> Atranspose:=Transpose(A); Btranspose:=Transpose(B);

Atranspose :=

⎡
⎣ 2 1 −2

1 −2 1
−1 3 2

⎤
⎦ Btranspose :=

⎡
⎣ 1 −2 2

−1 1 −1
2 3 1

⎤
⎦

Calculating (A B)T −BT AT in eq2 yields a null matrix, confirming the relation.

> eq2:=Transpose(A . B)-(Btranspose . Atranspose);

eq2 :=

⎡
⎣ 0 0 0

0 0 0
0 0 0

⎤
⎦

3.3.2 Eigenvalues and Eigenvectors

A man can become so accustomed to the thought of his own
faults that he will begin to cherish them as charming little
“personal characteristics”.
Helen Rowland, American journalist, A Guide to Men,“Brides” (1922)

Let A be an n × n matrix and X a column vector with n elements. The
equation A X =λX, or (A − λ I) X = 0, where λ is a number and I the unit or
identity matrix, has non-trivial solutions if and only if det(A − λ I)=0. When
expanded, the determinant yields a polynomial equation of degree n in λ, called
the characteristic polynomial equation. The roots of this polynomial equation
are called the characteristic values or eigenvalues of the matrix A. Correspond-
ing to each eigenvalue, there will be a non-trivial (X �= 0) solution X, which is
called a characteristic vector or eigenvector.

Find the eigenvalues and corresponding eigenvectors of the matrix

A :=

⎛
⎝ 5 7 −5

0 4 −1
2 8 −3

⎞
⎠ .

Determine a set of unit eigenvectors. A unit eigenvector has unit length, i.e.,
the sum of the squares of its components is 1.

To solve this problem, let’s load the LinearAlgebra package and enter
the given matrix A.

> restart: with(LinearAlgebra):

> A:=<<5|7|-5>,<0|4|-1>,<2|8|-3>>;

A :=

⎡
⎣ 5 7 −5

0 4 −1
2 8 −3

⎤
⎦

This problem can be tackled in two equivalent ways, either by mimicking the
hand calculation or using specific Maple commands which accomplish the same
task. Both ways will be demonstrated. The 3 × 3 identity matrix is entered.

> IM:=IdentityMatrix(3);

IM :=

⎡
⎣ 1 0 0

0 1 0
0 0 1

⎤
⎦

The identity matrix is multiplied by λ in S. The simplify command must be
applied to actually accomplish the multiplication. The same result could be
alternatively achieved by entering ScalarMatrix(lambda,3,3).

> S:=simplify(lambda*IM);

S :=

⎡
⎣ λ 0 0

0 λ 0
0 0 λ

⎤
⎦

If we were proceeding by hand, the characteristic matrix A − λ I would be
formed. This is illustrated in CM .

> CM:=A-S;

CM :=

⎡
⎣ 5 − λ 7 −5

0 4 − λ −1
2 8 −3 − λ

⎤
⎦

The characteristic matrix may be more simply obtained directly from A by
using the CharacteristicMatrix command as in CM2 .

> CM2:=CharacteristicMatrix(A,lambda);

CM2 :=

⎡
⎣ 5 − λ 7 −5

0 4 − λ −1
2 8 −3 − λ

⎤
⎦

Proceeding by hand, we then would determine the characteristic polynomial
equation by taking the determinant of the characteristic matrix and setting it
equal to zero. This is done in CP , a cubic polynomial in λ resulting.

> CP:=Determinant(CM)=0;

CP := 6 − 11 λ + 6 λ2 − λ3 = 0
The same equation follows on applying the CharacteristicPolynomial com-
mand directly to A and setting the result equal to zero.

> CP2:=CharacteristicPolynomial(A,lambda)=0;

CP2 := λ3 − 6 λ2 + 11 λ − 6 = 0
In the hand calculation, the eigenvalues are obtained by solving the character-
istic polynomial equation CP for λ, 3 eigenvalues resulting in L.

> L:=solve(CP,lambda);

L := 1, 2, 3
The same three eigenvalues again are more simply obtained by applying the
Eigenvalues command to A. Unless otherwise specified, the eigenvalues appear
by default in a column format.

> Eigenvalues(A); ⎡
⎣ 1

2
3

⎤
⎦

Instead of a column format, including the option output=list will produce the
eigenvalues in a list format as shown in the following command line.

> Eigenvalues(A,output=list);

[1, 2, 3]
Now that the eigenvalues are determined, the corresponding eigenvectors will
be calculated. If proceeding by hand, the eigenvalues are extracted separately
from L and labeled L1 , L2 , and L3 .

> L1:=L[1]; L2:=L[2]; L3:=L[3];

L1 := 1 L2 := 2 L3 := 3
A general eigenvector X, with three elements, is then entered as a column
vector. Note that the elements are separated by commas when inputting a
column vector. The goal is to determine the value of x1, x2, and x3 for each of
the eigenvalues.

> X:=<<x1,x2,x3>>;

X :=

⎡
⎣ x1

x2
x3

⎤
⎦

Then the column vector X is multiplied by the characteristic matrix CM , the
latter being evaluated for a specific eigenvalue, e.g., λ=L1 .

> eq1:=eval(CM,lambda=L1) . X;

eq1 :=

⎡
⎣ 4 x1 + 7 x2 − 5 x3

3 x2 − x3
2 x1 + 8 x2 − 4 x3

⎤
⎦

Extracting each row entry from eq1 and setting it equal to 0 yields a set of
equations in system1 for x1, x2, and x3.

> system1:={eq1[1,1]=0,eq1[2,1]=0,eq1[3,1]=0};
system1 := {4 x1 + 7 x2 − 5 x3 = 0, 3 x2 − x3 = 0, 2 x1 + 8 x2 − 4 x3 = 0}

The system of equations is solved in sol1 for the unknowns, the answer being
expressed in terms of the arbitrary quantity x2.

> sol1:=solve(system1,{x1,x2,x3});
sol1 := {x3 = 3 x2 , x1 = 2 x2 , x2 = x2}

The eigenvector X1 corresponding to the eigenvalue L1 is now obtained by
evaluating X with sol1 .

> X1:=eval(X,sol1);

X1 :=

⎡
⎣ 2 x2

x2
3 x2

⎤
⎦

There is an infinite family of eigenvectors X1 depending on the value chosen for
x2 . Proceeding by hand, the other two eigenvectors could be similarly obtained.
The choice of a specific value for x2 will be postponed for the moment.

The multitude of steps in deriving the three eigenvalues and eigenvectors
can be completely bypassed by applying the Eigenvectors command to A.

> (eiv,V):=Eigenvectors(A);

eiv , V :=

⎡
⎣ 1

2
3

⎤
⎦ ,

⎡
⎢⎢⎢⎢⎣

2
3

1
2

−1

1
3

1
2

1

1 1 1

⎤
⎥⎥⎥⎥⎦

In the output, the eigenvalues are presented in column format (the order of the
entries may vary from one execution of the worksheet to the next) before the
comma. Corresponding to the first (top) entry, the corresponding eigenvector
is given by the first column in the matrix format after the comma. The second
column corresponds to the second eigenvalue entry, and so on. So that the eigen-
values and eigenvectors can be extracted separately, the assignment (eiv,V) is
made, eiv corresponding to the eigenvalues and V to the eigenvectors.

Choosing x2=1/3, the “hand calculated” eigenvector X1 , corresponding to
the eigenvalue 1, is identical to the first column in V .

> X1:=eval(X1,x2=1/3);

X1 :=

⎡
⎢⎢⎢⎢⎣

2
3
1
3
1

⎤
⎥⎥⎥⎥⎦

The three eigenvectors are now extracted from V by using the Column command.
The second argument specifies the column to be extracted.

> V1:=Column(V,1); V2:=Column(V,2); V3:=Column(V,3);

V1 :=

⎡
⎢⎢⎢⎢⎣

2
3
1
3
1

⎤
⎥⎥⎥⎥⎦ V2 :=

⎡
⎢⎢⎢⎢⎣

1
2
1
2
1

⎤
⎥⎥⎥⎥⎦ V3 :=

⎡
⎣ −1

1
1

⎤
⎦

Finally, it was requested that a set of unit eigenvectors be obtained. The sum
of the squares of the elements of each eigenvector can be obtained by applying
the Norm(V,2) command. Dividing each of the above eigenvectors by its norm,
produces a unit eigenvector.

> U1:=V1/Norm(V1,2); U2:=V2/Norm(V2,2); U3:=V3/Norm(V3,2);

U1 :=

⎡
⎢⎢⎢⎢⎢⎢⎢⎣

√
14
7√
14

14
3

√
14

14

⎤
⎥⎥⎥⎥⎥⎥⎥⎦

U2 :=

⎡
⎢⎢⎢⎢⎢⎢⎢⎣

√
6

6√
6

6√
6

3

⎤
⎥⎥⎥⎥⎥⎥⎥⎦

U3 :=

⎡
⎢⎢⎢⎢⎢⎢⎢⎣

−
√

3
3√
3

3√
3

3

⎤
⎥⎥⎥⎥⎥⎥⎥⎦

When first learning how to find eigenvalues and eigenvectors, its important to
go through the steps outlined in “our hand calculation”. But, after doing this
a few times, you will be grateful for Maple’s specialized commands.

3.3.3 Diagonalizing a Matrix

An idea is a point of departure and no more. As soon as you
elaborate it, it becomes transformed by thought.
Pablo Picasso, Spanish artist, (1881–1973)

Find a matrix C which reduces the matrix

A =

⎛
⎝ 2 4 −6

4 2 −6
−6 −6 −15

⎞
⎠

to diagonal form (all elements not on the main diagonal are zero) by the trans-
formation C−1 A C.

After loading the LinearAlgebra package, the matrix A is entered.
> restart: with(LinearAlgebra):

> A:=<<2|4|-6>,<4|2|-6>,<-6|-6|-15>>;

A :=

⎡
⎣ 2 4 −6

4 2 −6
−6 −6 −15

⎤
⎦

A general matrix C with doubly subscripted elements is entered.
> C:=<<c[1,1]|c[1,2]|c[1,3]>,<c[2,1]|c[2,2]|c[2,3]>,

<c[3,1]|c[3,2]|c[3,3]>>;

C :=

⎡
⎣ c1, 1 c1, 2 c1, 3

c2, 1 c2, 2 c2, 3
c3, 1 c3, 2 c3, 3

⎤
⎦

The eigenvalues of A are obtained.
> L:=Eigenvalues(A);

L :=

⎡
⎣ −2

9
−18

⎤
⎦

L gives the three distinct eigenvalues. It is desired to find a matrix C which
reduces A to a diagonal matrix S with the main diagonal elements given by the
eigenvalues and the off-diagonal elements equal to zero. The DiagonalMatrix
command is used to form S.

> S:=DiagonalMatrix([L[1],L[2],L[3]],3,3);

S :=

⎡
⎣ −2 0 0

0 9 0
0 0 −18

⎤
⎦

Now, we want C−1 A C =S. Multiplying this matrix equation from the left by
C and noting that C C−1 = I, we have I A C = A C = C S, or A C − C S = 0.
The left-hand side of this last matrix equation is now entered in E.

> E:=(A . C)-(C . S);

E :=
[
4 c1, 1 + 4 c2, 1 − 6 c3, 1 , −7 c1, 2 + 4 c2, 2 − 6 c3, 2 , 20 c1, 3 + 4 c2, 3 − 6 c3, 3

]
[
4 c1, 1 + 4 c2, 1 − 6 c3, 1 , 4 c1, 2 − 7 c2, 2 − 6 c3, 2 , 4 c1, 3 + 20 c2, 3 − 6 c3, 3

]
[
− 6 c1, 1 − 6 c2, 1 − 13 c3, 1 , −6 c1, 2 − 6 c2, 2 − 24 c3, 2 ,−6 c1, 3 − 6 c2, 3 + 3 c3, 3

]
We have to solve the system of equations resulting from setting the 9 entries
of E equal to zero. This system of 9 equations is now obtained by using the
sequence command twice to sum over the rows and columns of E.

> System:={seq(seq(E[i,j]=0,j=1..3),i=1..3)};
System := {4 c1, 1 + 4 c2, 1 − 6 c3, 1 = 0, −7 c1, 2 + 4 c2, 2 − 6 c3, 2 = 0,

20 c1, 3 + 4 c2, 3 − 6 c3, 3 = 0, 4 c1, 2 − 7 c2, 2 − 6 c3, 2 = 0,

4 c1, 3 + 20 c2, 3 − 6 c3, 3 = 0, −6 c1, 1 − 6 c2, 1 − 13 c3, 1 = 0,

−6 c1, 2 − 6 c2, 2 − 24 c3, 2 = 0, −6 c1, 3 − 6 c2, 3 + 3 c3, 3 = 0}
The double sequence command is used again to enter the 9 unknown matrix
elements to be solved for.

> Elements:={seq(seq(c[i,j],j=1..3),i=1..3)};
Elements := {c1, 1, c1, 2, c1, 3, c2, 1, c2, 2, c2, 3, c3, 1, c3, 2, c3, 3}

The system of 9 equations is solved for the 9 unknown matrix elements.

> Solution:=solve(System,Elements);

Solution := {c1, 3 = c2, 3, c3, 3 = 4 c2, 3, c1, 2 = −2 c3, 2, c1, 1 = −c2, 1,

c2, 1 = c2, 1, c2, 3 = c2, 3, c3, 2 = c3, 2, c3, 1 = 0, c2, 2 = −2 c3, 2}
Then Solution is used to evaluate C.

> C:=eval(C,Solution);

C :=

⎡
⎣ −c2, 1 −2 c3, 2 c2, 3

c2, 1 −2 c3, 2 c2, 3
0 c3, 2 4 c2, 3

⎤
⎦

The matrix elements c2,1, c3,2, and c2,3 are undetermined. We are free to
choose values for these elements. There are an infinity of choices and therefore
of C matrices which will diagonalize A with the transformation C−1 A C. Let’s
check that this is so, keeping the three elements unspecified for the moment.
The inverse matrix C−1 is calculated.

> Cinverse:=MatrixInverse(C);

Cinverse :=

⎡
⎢⎢⎢⎢⎢⎢⎣

−1
2

1
c2, 1

1
2

1
c2, 1

0

−2
9

1
c3, 2

−2
9

1
c3, 2

1
9

1
c3, 2

1
18

1
c2, 3

1
18

1
c2, 3

2
9

1
c2, 3

⎤
⎥⎥⎥⎥⎥⎥⎦

Provided that non-zero values are chosen for the undetermined elements, the
inverse matrix C−1 exists. Now we calculate C−1 A C, obtaining the diagonal
matrix S as expected.

> check:=Cinverse . (A . C);

check :=

⎡
⎣ −2 0 0

0 9 0
0 0 −18

⎤
⎦

A particular matrix C which will accomplish the transformation is now obtained
by setting all the undetermined matrix elements equal to 1.

> C:=eval(C,{c[2,1]=1,c[3,2]=1,c[2,3]=1});

C :=

⎡
⎣ −1 −2 1

1 −2 1
0 1 4

⎤
⎦

3.3.4 Orthogonal and Unitary Matrices

No matter how calmly you try to referee, parenting will eventually
produce bizarre behavior, and I’m not talking about the kids. Their
behavior is always normal.
Bill Cosby, American comedian, Fatherhood, 1986

The ideas of orthogonality and normalization which are important in the discus-
sion of vectors can be extended to matrices as illustrated in this simple recipe.

(a) If A =

⎛
⎝ a1

a2
a3

⎞
⎠ , B =

⎛
⎝ b1

b2
b3

⎞
⎠ , C =

⎛
⎝ c1

c2
c3

⎞
⎠

are real mutually orthogonal unit column vectors, prove that the real matrix

ABC =

⎛
⎝ a1 b1 c1

a2 b2 c2
a3 b3 c3

⎞
⎠

is an orthogonal matrix. A real matrix M is referred to as an orthogonal matrix
if its transpose is equal to its inverse, i.e., MT =M−1, or MT M =I.

(b) With i=
√−1, show that

H =

⎛
⎜⎜⎜⎜⎜⎝

√
2

2
−i

√
2

2
0

i

√
2

2
−

√
2

2
0

0 0 1

⎞
⎟⎟⎟⎟⎟⎠

is a unitary matrix. A complex matrix M is a unitary matrix if its complex
conjugate transpose is equal to its inverse, i.e., (M∗)T =M−1, or (M∗)T M =I.

After the usual loading of the LinearAlgebra package when dealing with ma-
trices, the column vectors A, B, and C are entered using the Vector command,
each argument containing a list of the relevant elements.

> restart: with(LinearAlgebra):

> A:=Vector([a1,a2,a3]); B:=Vector([b1,b2,b3]);

C:=Vector([c1,c2,c3]);

A :=

⎡
⎣ a1

a2
a3

⎤
⎦ B :=

⎡
⎣ b1

b2
b3

⎤
⎦ C :=

⎡
⎣ c1

c2
c3

⎤
⎦

A functional operator F is introduced for imposing “orthonormality” on arbi-
trary column vectors M and N . Normalization follows on setting the parameter
n=1, orthogonality on setting n=0.

> F:=(M,N,n)->(Transpose(M) . N)=n:

Using F, normalization is imposed on A, B, and C in eq1 , eq2 , and eq3 , and
orthogonality in eq4 , eq5 , and eq6 .

> eq||1:=F(A,A,1); eq||2:=F(B,B,1); eq||3:=F(C,C,1);

eq1 := a1 2 + a2 2 + a3 2 = 1 eq2 := b1 2 + b2 2 + b3 2 = 1

eq3 := c1 2 + c2 2 + c3 2 = 1
> eq||4:=F(A,B,0); eq||5:=F(A,C,0); eq||6:=F(B,C,0);

eq4 := b1 a1 + b2 a2 + b3 a3 = 0 eq5 := c1 a1 + c2 a2 + c3 a3 = 0
eq6 := c1 b1 + c2 b2 + c3 b3 = 0

The matrix ABC is entered using the Matrix command.
> ABC:=Matrix([[a1,b1,c1],[a2,b2,c2],[a3,b3,c3]]);

ABC :=

⎡
⎣ a1 b1 c1

a2 b2 c2
a3 b3 c3

⎤
⎦

To produce an orthogonal matrix OM , the product (ABC )T (ABC ) is formed.
> OM:=Transpose(ABC) . ABC;

OM :=

⎡
⎣ a1 2 + a2 2 + a3 2 , b1 a1 + b2 a2 + b3 a3 , c1 a1 + c2 a2 + c3 a3

b1 a1 + b2 a2 + b3 a3 , b1 2 + b2 2 + b3 2 , c1 b1 + c2 b2 + c3 b3
c1 a1 + c2 a2 + c3 a3 , c1 b1 + c2 b2 + c3 b3 , c1 2 + c2 2 + c3 2

⎤
⎦

Evaluating OM with the six orthonormality conditions, yields the identity
(unit) matrix, thus confirming that ABC is an orthogonal matrix.

> OM2:=eval(OM,{seq(eq||i,i=1..6)});

OM2 :=

⎡
⎣ 1 0 0

0 1 0
0 0 1

⎤
⎦

To answer part (b), the interface(imaginaryunit=i) command is used to
set i ≡ √−1. Then the matrix H is entered.

> interface(imaginaryunit=i):

> H:=<<sqrt(2)/2|-i*sqrt(2)/2|0>,<i*sqrt(2)/2|-sqrt(2)/2|0>,

<0|0|1>>;

H :=

⎡
⎢⎢⎢⎢⎢⎣

√
2

2
−1
2

i
√

2 0

1
2

i
√

2 −
√

2
2

0

0 0 1

⎤
⎥⎥⎥⎥⎥⎦

The HermitianTranspose command is applied to H in T . This command
applies the transpose (interchanging rows and columns) and takes the complex
conjugate of the elements.

> T:=HermitianTranspose(H);

T :=

⎡
⎢⎢⎢⎢⎢⎣

√
2

2
−1
2

i
√

2 0

1
2

i
√

2 −
√

2
2

0

0 0 1

⎤
⎥⎥⎥⎥⎥⎦

Since T = H, H is a Hermitian matrix. That it is also a unitary matrix is
confirmed by forming the product H T , the result R being the unit matrix.

> R:=H . T;

R :=

⎡
⎣ 1 0 0

0 1 0
0 0 1

⎤
⎦

3.3.5 Introducing the Euler Angles

To knock a thing down, especially if it is cocked at an arrogant angle,
is a deep delight to the blood.
George Santayana, American philosopher (1863–1952)

The transformation from one 3-dimensional coordinate system (x, y, z) to an-
other (x ′,y ′,z ′ ) with the same origin can be represented by a matrix equation
of the form X ′ =R X, where R is the 3 × 3 rotation matrix and

X =

⎛
⎝ x

y
z

⎞
⎠ , X ′ =

⎛
⎝ x ′

y ′

z ′

⎞
⎠ .

The over-all rotation is made up of three 1-dimensional rotations through the
angles φ, θ, and ψ, called the Euler angles. The choices of these angles is not
uniform in the literature4, so I will adopt the notation of Marion and Thornton
[MT95] and Goldstein, Poole, and Safco [GPS02].

4In fact, Euler angles can be avoided altogether in classical physics by specifying a single
plane of rotation and one angle, as discussed for example in [Bay99], [DL03], and [Hes99].

The first rotation is counterclockwise through an angle φ about the z axis,
the rotation matrix being

Rφ =

⎛
⎝ cos φ sin φ 0

− sin φ cos φ 0
0 0 1

⎞
⎠ .

The new axes are labeled x1, y1, and z1 = z. The second rotation is coun-
terclockwise through an angle θ about the x1 axis, the rotation matrix being

Rθ =

⎛
⎝ 1 0 0

0 cos θ sin θ
0 − sin θ cos θ

⎞
⎠ .

The new axes are labeled x2 =x1, y2, z2. The third rotation is counterclockwise
through an angle ψ about the z2 axis, the rotation matrix being

Rφ =

⎛
⎝ cos ψ sin ψ 0

− sin ψ cos ψ 0
0 0 1

⎞
⎠ .

The new axes are labeled x ′, y ′, z ′ = z2. The total rotation matrix then is
R=Rψ Rθ Rφ.

In this recipe, I will derive the total rotation matrix R and apply its inverse
to a triplet of vectors originally pointing along the x, y, z axes. The plots and
LinearAlgebra packages are first loaded.

> restart: with(plots): with(LinearAlgebra):

To ease the typing, the aliases p, t, and s are introduced for the input quantities
phi, theta, and psi.

> alias(phi=p,theta=t,psi=s):

The three rotation matrices Rφ, Rθ, and Rψ are entered.
> R[p]:=<<cos(p)|sin(p)|0>,<-sin(p)|cos(p)|0>,<0|0|1>>;

Rφ :=

⎡
⎣ cos(φ) sin(φ) 0

−sin(φ) cos(φ) 0
0 0 1

⎤
⎦

> R[t]:=<<1|0|0>,<0|cos(t)|sin(t)>,<0|-sin(t)|cos(t)>>;

Rθ :=

⎡
⎣ 1 0 0

0 cos(θ) sin(θ)
0 −sin(θ) cos(θ)

⎤
⎦

> R[s]:=<<cos(s)|sin(s)|0>,<-sin(s)|cos(s)|0>,<0|0|1>>;

Rψ :=

⎡
⎣ cos(ψ) sin(ψ) 0

−sin(ψ) cos(ψ) 0
0 0 1

⎤
⎦

The total rotation matrix R=Rψ Rθ Rφ is then calculated.
> R:=R[s] . (R[t] . R[p]);

R :=

⎡
⎣ cosψ cosφ − sinψ cosθ sinφ, cosψ sinφ + sinψ cosθ cosφ, sinψ sinθ

−sinψ cosφ − cosψ cosθ sinφ, −sinψ sinφ + cosψ cosθ cosφ, cosψ sinθ
sinθ sinφ, −sinθ cosφ, cosθ

⎤
⎦

The inverse rotation matrix Rinv for rotating from the primed coordinates to
the unprimed ones is obtained by applying the MatrixInverse command to R
and simplifying with the trigonometric option.

> Rinv:=simplify(MatrixInverse(R),trig);

Rinv :=

⎡
⎣ cosψ cosφ − sinψ cosθ sinφ, −sinψ cosφ − cosψ cosθ sinφ, sinθ sinφ

cosψ sinφ + sinψ cosθ cosφ, −sinψ sinφ + cosψ cosθ cosφ, −sinθ cosφ
sinψ sinθ, cosψ sinθ, cosθ

⎤
⎦

To illustrate vector rotation (the inverse of coordinate rotation), let’s evaluate
Rinv for φ=π/3, θ= π/4, and ψ=π/6 radians.

> Rinv:=eval(Rinv,{p=Pi/3,t=Pi/4,s=Pi/6});

Rinv :=

⎡
⎢⎢⎢⎢⎢⎢⎢⎣

√
3

4
−

√
2

√
3

8
−1

4
− 3

√
2

8

√
2

√
3

4
3
4

+
√

2
8

−
√

3
4

+
√

2
√

3
8

−
√

2
4√

2
4

√
2

√
3

4

√
2

2

⎤
⎥⎥⎥⎥⎥⎥⎥⎦

The following column vectors, each of length 8 units, are entered.
> v1:=<8,0,0>: v2:=<0,8,0>: v3:=<0,0,8>:

Then the inverse rotation matrix Rinv is applied to each vector.
> w1:=Rinv . v1; w2:=Rinv . v2; w3:=Rinv . v3;

w1 :=

⎡
⎣ 2

√
3 − √

2
√

3
6 +

√
2

2
√

2

⎤
⎦ w2 :=

⎡
⎣ −2 − 3

√
2

−2
√

3 +
√

2
√

3
2

√
2

√
3

⎤
⎦ w3 :=

⎡
⎣ 2

√
2

√
3

−2
√

2
4

√
2

⎤
⎦

The rotated vectors will be of exactly the same length as the original vectors.
For example, it is confirmed in the following command line, using the Norm
command, that w1 is 8 units long.

> simplify(Norm(w1,2));

8
To plot the original and rotated vectors, an arrow operator F is formed to plot
an arbitrary vector v as a cylindrical arrow with its tail at the origin. The color
c of the arrow must be specified. Using F, then v1 , v2 , and v3 are plotted as
red arrows, and w1 , w2 , and w3 as blue arrows.

> F:=(v,c)->arrow(<0,0,0>,v,shape=cylindrical arrow,color=c,

width=0.15,head width=0.4,head length=0.5):

> a[1]:=F(v1,red): a[2]:=F(v2,red): a[3]:=F(v3,red):

> a[4]:=F(w1,blue): a[5]:=F(w2,blue): a[6]:=F(w3,blue):



122 CHAPTER 3. VECTORS AND MATRICES

The six arrows are superimposed with the display command with constrained
scaling and a specified orientation of the viewing box, the resulting picture
being shown in Figure 3.7. The original vectors v1 , v2 , and v3 are the ones
oriented along the coordinate axes.

> display({seq(a[i],i=1..6)},axes=normal,
orientation=[-36,80],scaling=constrained);

2

4

6

8

–2
4 6 8–6 –4

2 4 6 8

Figure 3.7: Matrix rotation of three vectors.

3.4 Supplementary Recipes
03-S01: Bobby Blowfly Seeks a Warmer Clime
The temperature in a certain region is given by T = (x−3)2+(y−2)2+3 (z−1)2,
with distances in meters and temperatures in degrees Celsius. Bobby Blowfly
is currently at the point x = 1, y = 2, z = 3. What is the temperature at this
point? Bobby wishes to start flying in a direction which produces the maximum
rate of temperature increase. What is the unit vector r̂ which points in this
direction? What is the maximum rate of temperature change at (1, 2, 3)? How
much greater is this rate than if he had decided to fly in the direction given
by the unit vector û = (êx − êy + êz)/

√
3? Plot the 3-dimensional isothermal

(constant temperature) surface passing through the point (1, 2, 3) along with
the unit vectors r̂ and û with their tails placed at the point.

03-S02: Jennifer’s Vector Assignment
Jennifer asks you to try the following problem. Consider the vectors
�A = 3 êx − êy − 7 êz, �B = 4 êx + êy + 2 êz, and �C = êx + 5 êy + 4 êz.

(a) Plot the three vectors as cylindrical arrows in a 3-dimensional picture.

(b) Calculate �P1 =( �A × �B) × �C and �P2 = �A × ( �B × �C). Confirm �P1 �= �P2.

(c) Calculate the volume V = | �A · ( �B × �C)| of a parallelepiped with sides �A,

�B, and �C. Show that V =

∣∣∣∣∣∣
Ax Ay Az

Bx By Bz

Cx Cy Cz

∣∣∣∣∣∣ .

03-S03: Another Vector Operator Identity
Using spherical polar coordinates, prove the following vector identity for gen-
eral 3-dimensional vector fields �A and �B:

∇ · ( �A × �B) = �B · (∇ × �A) − �A · (∇ × �B).
03-S04: Another Maple Approach
Using the LinearAlgebra package instead of the VectorCalculus package, derive
the velocity and acceleration of a particle in spherical polar coordinates.

03-S05: Conservative or Non-conservative?
Consider the vector field

�A = (3x2 − 6 y z) êx + (2 y + 3 x z) êy + (1 − 4 x y z2) êz.

(a) Calculate the line integral of �A along the straight line joining the points
(0, 0, 0) and (1, 1, 1).

(b) Calculate the line integral of �A along the straight lines from (0, 0, 0) to
(0, 0, 1), then to (0, 1, 1), and then to (1, 1, 1).

(c) Calculate the line integral of �A from (0, 0, 0) to (1, 1, 1) along the path
x= t, y= t2, z= t3, where t is a parameter.

(d) Having performed the line integrals, is the vector field �A conservative or
non-conservative? Confirm your conclusion by calculating the curl of �A.

(e) Make a 3-dimensional plot of the vector field over an appropriate range of
x, y, and z. Is the distribution of arrows consistent with your conclusion?

03-S06: Basic Matrix Operations

Consider the two square matrices, A =

⎛
⎝ 1 0 2

2 −1 3
0 5 6

⎞
⎠ , B =

⎛
⎝ 2 −1 1

0 1 2
1 −2 −1

⎞
⎠.

(a) Use the Dimension command to confirm that A and B are 3 × 3 matrices.
(b) By calculating their determinants, show that A and B are non-singular.
(c) Calculate A + B, A − B, and A2 − B2.
(d) Show that A B �= B A, i.e., the two matrices do not commute.

(e) Introducing the column vector C =

⎛
⎝ 2

1
3

⎞
⎠, calculate (A B) C.

(f) Show that the product C (A B) is invalid.
(g) Convert C to a row vector C2 by using the Transpose command.
(h) Calculate the product C2 (A B).

03-S07: The Cayley–Hamilton Theorem
Given the 4 × 4 matrix

A =

⎛
⎜⎜⎝

2 1 −1 1
−2 1 3 −2

1 −2 1 1
−3 1 2 −1

⎞
⎟⎟⎠ :

(a) Determine the eigenvalues and eigenvectors of A.

(b) Calculate the determinant and trace of A.

(c) Evaluate A2, A3, and A4. Using these results, verify the Cayley–Hamilton
theorem, which states that a matrix satisfies its own characteristic poly-
nomial equation.

03-S08: Simultaneous Diagonalization
Consider the following 4 × 4 matrices,

A =

⎛
⎜⎜⎝

1 3 0 0
3 1 0 0
0 0 1 2
0 0 2 1

⎞
⎟⎟⎠ , B =

⎛
⎜⎜⎝

2 3 0 0
3 2 0 0
0 0 2 1
0 0 1 2

⎞
⎟⎟⎠ .

(a) Show that A B=B A. The matrices are said to commute.

(b) Determine the eigenvalues of A. Find a general matrix C that reduces A
to diagonal form by the transformation C−1 A C.

(c) Determine the eigenvalues of B. Show that the matrix C of part (b) also
reduces B to diagonal form by the transformation C−1 B C.

03-S09: Orthonormal Vectors
Show that the following vectors form an orthonormal set.

A1 =

⎛
⎝ cos(θ)

sin(θ)
0

⎞
⎠ , A2 =

⎛
⎝ −sin(θ)

cos(θ)
0

⎞
⎠ , A3 =

⎛
⎝ 0

0
1

⎞
⎠ .

03-S10: Stokes’s Theorem
Stokes’s theorem for a vector field �A states that

∮
C

�A · d�� =
∫

S
(∇ × �A) · d�a,

where d�� is an element of vector path length along the closed contour C and d�a
is an element of area on the open surface S bounded by C. The direction of the
vector area element is related to the direction of the line integral in a right-hand
sense. Curling the fingers of the right hand in the direction of performing the
line integral, the thumb indicates the sense of the vector area.

Verify Stokes’s theorem for �A=3 y êx−x z êy +y z2 êz, where S is the surface
of the paraboloid 2 z = x2 + y2 bounded by z=2 and C is its boundary. I.e., C
is the circle x2 + y2 = 4 lying in the x-y plane at z=2.

03-S11: Solving Linear Equation Systems
The currents I1, I2, I3, I4 in an electrical network satisfy the system of equations

3 I1 + 2 I2 − I4 = 65, 2 I1 − I2 + 4 I3 + 3 I4 = 160,

−7 I1 − 4 I2 − 2 I4 = 23, 5 I1 − I2 − 2 I3 + I4 = 3.

Write the system in matrix form and then determine the currents by (a) using an
inverse matrix approach (b) using the direct approach with the solve command.



Part II

THE ENTREES

The mathematics is not there till we put it there.
Sir Arthur Eddington, The Philosophy of Physical Science, (1882 – 1944)

If scientific reasoning were limited to the logical
processes of arithmetic, we should not get very far

in our understanding of the physical world.
One might as well attempt to grasp the game of poker
entirely by the use of the mathematics of probability.

Vannevar Bush, American scientist (1890 – 1974)

A man ceases to be a beginner in any given science
and becomes a master in that science when he has

learned that...he is going to be a beginner all his life.
R. G. Collingwood, British philosopher (1889–1943)

I believe that a scientist looking at nonscientific
problems is just as dumb as the next guy.

Richard Feynman, American physicist (1918–1988)



Chapter 4

Linear PDEs of Physics

In this chapter, we will solve a wide variety of physical problems involving
linear PDEs, first in Cartesian coordinates, then in other curvilinear coordinate
systems. The common linear PDEs of physics include:

(1) The wave equation (with ψ a continuous scalar function of position �r and
time t, and c the wave velocity),

∇2ψ =
1
c2

∂2ψ

∂t2
, (4.1)

for vibrating elastic strings and membranes, sound waves in fluids, elastic
waves in solids, electromagnetic waves (with ψ a vector), etc.

(2) The diffusion equation (with d a positive diffusion constant),

∂ψ

∂t
= d ∇2ψ, (4.2)

which applies to time-dependent heat flow, mixing of fluids, neutron dif-
fusion in nuclear reactors, diffusion of impurities in solids, and so on.

(3) Laplace’s equation,
∇2ψ = 0, (4.3)

which applies to steady-state heat flow, irrotational flow of an incompress-
ible fluid, and electro(magneto)statics in charge(current)-free regions.

(4) Poisson’s equation (with S a source term)

∇2ψ = S(�r, t), (4.4)

which applies, for example, to the electrostatic potential due to a charge
distribution and to the steady-state temperature with a heat source present.

(5) The time-independent Schrödinger equation (with h̄ = h/2π (h being
Planck’s constant), m the mass, V the potential, and E the total energy),

− h̄2

2m
∇2ψ + V ψ = E ψ, (4.5)

which describes stationary states in quantum mechanics.

4.1 Three Cheers for the String

When I introduce my physics and engineering students to the PDEs of physics,
I always begin with the transverse vibrations of the humble string. The reason
is quite simple. The vibrating string is familiar and easy to visualize, the basic
underlying equation of motion (the 1-dimensional wave equation) easy to derive,
and a wide variety of important methods and ideas can be introduced. So this
section illustrates a gourmet selection of string recipes. Lest you get ideas of
stringing me up after munching on some of the stringy concoctions, I will show
you that there is life beyond the string in the second part of this chapter.

4.1.1 Jennifer Finds the General Solution

Friends are like fiddle strings, they must not be screwed too tight.
English Proverb. Collected in: H. G. Bohn, A Handbook of Proverbs (1855)

Consider a light, uniform, stretched string of linear density (mass per unit
length) ε which is horizontal in equilibrium. The goal of this recipe, provided
by Jennifer the MIT mathematician, is to obtain the equation of motion for the
transverse (vertical) oscillations of the string and then the general solution.

The PDEtools package is loaded, because it contains the declare and
dchange commands. Jennifer needs the former to introduce subscript nota-
tion, favored by mathematicians, for the resulting 1-dimensional wave equation
and the latter to transform the variables in arriving at the general solution.
The plots package is included because a typical solution will be animated.

> restart: with(PDEtools): with(plots):

Let the vertical displacement of a point x on the string from equilibrium at
time t be ψ(x, t). Consider an infinitesimal element of string of arclength ds=√

(dx)2 + (dψ)2 =
√

1 + (∂ψ/∂x)2 dx, located between x and x + dx. Since the
string is only to move vertically, the horizontal component T of the tension in
the string is constant along the string. The vertical component of the tension,
which is given by T ∂ψ/∂x, will vary along the string. The net vertical force
F on the infinitesimal element is equal to the difference between the vertical
forces at its ends. This force is entered.

> F:=T*Diff(psi(x+dx,t),x)-T*Diff(psi(x,t),x);

F := T (
∂

∂x
ψ(x + dx , t)) − T (

∂

∂x
ψ(x, t))

Since dx is small, F is taylor expanded in powers of dx to order 2 and the order
of term removed with the convert( ,polynom) command.

> F:=convert(taylor(F,dx=0,2),polynom);

F := T D1(ψ)(x, t) − T (
∂

∂x
ψ(x, t)) + T D1, 1(ψ)(x, t) dx

Newton’s second law is applied to the string element in the vertical direction.
The net vertical force F is converted to differential form on the lhs of ode. This

must be equal to the mass of the string element times the acceleration, i.e., to
ε ds (∂2ψ/∂t2). Assuming that the vibrations are such that ∂ψ/∂x � 1, then
ds ≈ dx. This linear approximation is used on the rhs of ode.

> ode:=convert(F,diff)=epsilon*dx*diff(psi(x,t),t,t);

ode := T (
∂2

∂x2 ψ(x, t)) dx = ε dx (
∂2

∂t2
ψ(x, t))

Dividing ode by (dx T ) and substituting ε=T/c2, yields the 1-dimensional wave
equation WE , with c the wave speed. By using the declare command, Jennifer
has introduced subscript notation for the derivatives in the wave equation.

> declare(psi(x,t)): WE:=subs(epsilon=T/cˆ2,ode/(dx*T));

ψ(x, t)will now be displayed as ψ

WE := ψx, x =
ψt, t

c2

If proceeding by hand, the general solution of WE can be obtained by in-
troducing two new independent variables u and v related to x and t by the
transformation u=x + c t, v =x − c t. Then one would apply the chain rule of
differentiation, e.g., calculating,

∂ψ

∂x
=

∂ψ

∂u

∂u

∂x
+

∂ψ

∂v

∂v

∂x
=

∂ψ

∂u
+

∂ψ

∂v
, and then

∂2ψ

∂x2 =
∂2ψ

∂u2 + 2
∂2ψ

∂v∂u
+

∂2ψ

∂v2

The second time derivative would be similarly calculated. This hand trans-
formation of the wave equation WE can be easily accomplished with Maple
as follows. Jennifer enters the transformation tr and solves for x and t, thus
obtaining the inverse transformation itr .

> tr:={u=x+c*t,v=x-c*t}: itr:=solve(tr,{x,t});
itr := {t = −v − u

2 c
, x =

u

2
+

v

2
}

Then applying the inverse transformation to the wave equation with the dchange
command, and simplifying, yields the transformed wave equation WE2 . Note
that c is indicated as a parameter (params=c) in the command.

> WE2:=simplify(dchange(itr,WE,[u,v],params=c));

WE2 := ψu, u + 2 ψu, v + ψv, v = ψu, u − 2 ψu, v + ψv, v

The cross-derivative term is isolated in WE2 , yielding the greatly simplified
wave equation WE3 .

> WE3:=isolate(WE2,diff(psi(u,v),u,v));

WE3 := ψu, v = 0
Clearly, the solution of WE3 is a linear combination of an arbitrary function
of u and an arbitrary function of v. This general solution can be obtained by
applying the partial differential equation solve command, pdsolve, to WE3 .

> sol:=pdsolve(WE3);

sol := ψ(u, v) = F2 (u) + F1 (v)

The quantities F1 and F2 denote arbitrary functions. The general solution
to the original wave equation follows on substituting the transformation tr into
the right-hand side of the solution sol .

> sol2:=psi(x,t)=subs(tr,rhs(sol));

sol2 := ψ = F2 (x + c t) + F1 (x − c t)
Of course, since Jennifer is using Maple, it was not really necessary to mimic
the hand calculation. Applying the pdsolve command directly to the original
wave equation WE yields a similar general solution.

> sol3:=pdsolve(WE);

sol3 := ψ = F1 (x + c t) + F2 (c t − x)
The solution sol3 is a linear superposition of a general wave form traveling in the
negative-x direction (the first term) with speed c and a wave form (the second
term) traveling in the positive-x direction with the same speed. The wave forms
propagate unchanged in shape since dissipative forces have not been included.
As a representative example, Jennifer considers the following specific wave form
ψ which is a linear combination of two Gaussian profiles.

> psi:=exp(-(x-t)ˆ2)+exp(-(x+t)ˆ2);

ψ := e(−(x−t)2) + e(−(x+t)2)

On animating ψ with the following command, the initial frame shows a single
pulse of amplitude 2. As the animation progresses, this pulse splits into two
pulses of amplitude 1 propagating in opposite directions with speed c=1.

> animate(psi,x=-10..10,t=0..7,frames=50,numpoints=500);

4.1.2 Daniel Separates Strings: I Separate Variables

There are strings in the human heart that had better not be vibrated.
Charles Dickens, English novelist, Barnaby Rudge (1841)

In contrast to Dickens’s quote, there are strings in my heart which vibrate
with pleasure as I watch my young grandson, Daniel, explore his new world.
However, given a multi-stranded string, he would likely try to separate it into
its constituent strands, rather than carrying out the following scenario. But,
who knows? Perhaps, some day, he will separate variables instead of strings.

Young Daniel is playing with a light horizontal string of length L fixed at
x = 0 and L, i.e., ψ(0, t) = ψ(L, t) = 0. Suppose that he cleverly plucks the
string (which is initially at rest) in such a way that it has an initial profile
ψ(x, 0)=f(x)=h x3 (L−x)/L4. Our task is to use the method of separation of
variables to mathematically determine the subsequent motion of the string and
then animate the string vibrations, taking L = 1 m, h = 5 m, and c = 5 m/s.

After loading the plots package, needed for the animation,
> restart: with(plots):

the 1-dimensional wave equation is entered in pde. Unlike Jennifer, I will not
bother with the subscript notation.

> pde:=diff(psi(x,t),x,x)=(1/cˆ2)*diff(psi(x,t),t,t);

pde :=
∂2

∂x2 ψ(x, t) =

∂2

∂t2
ψ(x, t)

c2

The separation of variables method assumes that the solution can be written
as a product of unknown functions, each function depending on only one of
the independent variables. For the present problem, this assumption takes the
form ψ(x, t) = X(x) T (t). Mentally substituting ψ(x, t) into pde and dividing
by ψ(x, t) would yield (d2X(x)/dx2)/X(x)= (d2T (t)/dt2)/(c2 T (t)). The only
way this equality can hold is if both sides of the equation are equal to a con-
stant, called the separation constant. This then would yield two ODEs, one for
determining X(x) and a second for finding T (t). These steps may be achieved
by applying the pdsolve command to pde, but now including the product as-
sumption as a HINT.

> pdsolve(pde,psi(x,t),HINT=X(x)*T(t));

(ψ(x, t) = X (x)T (t)) &where [{ d2

dt2
T (t) = c1 T (t) c2,

d2

dx2 X (x) = c1 X (x)}]

The resulting ODEs are shown in the above output, with c1 the separation
constant. The next step is to solve the ODEs for X and T . If, in addition to
supplying the HINT, the INTEGRATE option is included, the separated ODEs are
readily solved with the pdsolve command.

> pdsolve(pde,psi(x,t),HINT=X(x)*T(t),INTEGRATE);

(ψ(x, t) = X (x)T (t)) &where[{{T (t) = C3 e(
√

c1 c t) + C4 e(−√
c1 c t)},

{X (x) = C1 e(
√

c1 x) + C2 e(−√
c1 x)}}]

Finally, the product of X and T must be formed to give ψ(x, t). Including the
build option in the pdsolve command accomplishes this step. In the recipes
which follow in this and the following chapter, all three options will be included
whenever we wish to separate variables. You will begin to really appreciate
the power of the pdsolve command when we tackle problems in non-Cartesian
coordinate systems.

> sol:=pdsolve(pde,psi(x,t),HINT=X(x)*T(t),INTEGRATE,build);

sol := ψ(x, t) = e(
√

c1 c t) C3 C1 e(
√

c1 x) +
e(

√
c1 c t) C3 C2
e(

√
c1 x)

+
C4 C1 e(

√
c1 x)

e(
√

c1 c t) +
C4 C2

e(
√

c1 c t) e(
√

c1 x)

To determine the four constants C1 , etc., the two boundary conditions and
the two initial conditions must be applied. In order to do this, the form of
the solution will be simplified somewhat. First, let’s make the substitution
c1 =−k2 on the rhs of sol and assume that k > 0.

> psi:=simplify(subs(_c[1]=-kˆ2,rhs(sol))) assuming k>0;

ψ := C3 C1 e(k (c t+x) I) + C3 C2 e(k (c t−x) I)

+ C4 C1 e(−I k (c t−x)) + C4 C2 e(−I k (c t+x))

Then ψ is converted to trigonometric form and the result expanded.

> psi2:=expand(convert(psi,trig));

The lengthy output, which has been suppressed here in the text, involves the
terms cos(k x), sin(k x), cos(c k t), and sin(c k t). To satisfy the boundary
condition at x = 0, the cos(k x) terms are removed from ψ2 by substituting
cos(k x)=0. At x=L, the boundary condition yields sin(k L)=0 or k=nπ/L,
with n = 1, 2, 3, .... This is also substituted. Since the transverse velocity of
the string is initially zero, we must also set sin(c k, t)=0.

> psi3:=subs({cos(k*x)=0,sin(c*k*t)=0,k=n*Pi/L},psi2);
On then factoring ψ3, the result shown in ψ4 results.

> psi4:=factor(psi3);

ψ4 := (− C2 + C1 ) ( C3 + C4 ) cos(
n π c t

L
) sin(

nπ x

L
) I

The following select command is used to replace the “ugly” coefficient com-
bination in ψ4 with the symbol An. The resulting term, labeled ψ5 here, is the
nth term in the infinite series which will represent ψ(x, t).

> psi5:=A[n]*select(has,psi4,{sin,cos});
ψ5 := An cos(

n π c t

L
) sin(

nπ x

L
)

The initial shape of the string is entered.

> f:=h*xˆ3*(L-x)/Lˆ4:

Using linear superposition, ψ(x, t) =
∑∞

n=1 An sin(n π x/L) cos(n π c t/L). At
t = 0, one has f =

∑∞
n=1 An sin(n π x/L). Multiplying this equation by

sin(m π x/L) and integrating from x=0 to L, the only term which will survive
in the series due to orthogonality of the sine functions is the term corresponding
to n=m. This procedure is carried out in the following equation.

> eq:=int(f*sin(n*Pi*x/L),x=0..L)

=int(subs(t=0,psi5)*sin(n*Pi*x/L),x=0..L):
The equation is then solved for An, assuming that n is an integer.

> A[n]:=solve(eq,A[n]) assuming n::integer;

An := −12 h (4 − 4 (−1)n + n2 π2 (−1)n)
n5 π5

The parameter values L = 1, h = 5, and c = 5 are entered and a formal sum
performed with the first 25 terms in the series being retained.

> L:=1: h:=5: c:=5:

> psi6:=Sum(psi5,n=1..25);

ψ6 :=
25∑

n=1

(−60 (4 − 4 (−1)n + n2 π2 (−1)n) cos(5 nπ t) sin(nπ x)
n5 π5 )

Applying the value command to ψ6, the motion of the string is now animated
over the time range t=0 to 160 seconds. The opening frame of the animation
is shown in Figure 4.1. By executing the command line, you will be able to
observe the vibrations of the string.

> animate(value(psi6),x=0..L,t=0..160,frames=100,

thickness=2,scaling=constrained,tickmarks=[4,3]);

–0.5

0

0.5

0.2 0.4 0.6 0.8 1
x

Figure 4.1: The initial shape of the plucked string.

4.1.3 Daniel Strikes Again: Mr. Fourier Reappears

We are all instruments endowed with feeling and memory.
Our senses are so many strings that are struck by surrounding
objects and that also frequently strike themselves.
Denis Diderot, French philosopher, (1713–84)

If Daniel didn’t separate the strands or otherwise destroy the string, another
possible scenario is that he would strike it in some manner. Consider a string
fixed at x = 0 and L to be initially horizontal, but is struck in such a way
that it is given the initial piecewise velocity profile g(x)=4 v x/L for x ≤ L/4,
g(x)= 4 (v/L) (L/2 − x) for L/4 ≤ x ≤ L/2, and g(x)= 0 for x ≥ L/2. Using
the Fourier series approach, determine the subsequent motion of the string and
animate it for L=20 cm, v=5 cm/s, and c=1 cm/s.

The plots package is needed for the animation. To perform the integration
involving the piece-wise velocity profile, it is necessary to assume that both v
and L are positive.

> restart: with(plots): assume(v>0,L>0):

A functional operator ψ is introduced to generate the necessary Fourier series.
From the previous recipe, it is clear that the spatial part should be built up
of terms of the form sin(n π x/L), with n = 1, 2, 3, ..., in order to satisfy the
boundary conditions at x=0 and L. The time part is left quite general, so that
the recipe will run even if the initial conditions are changed.

> psi:=N->Sum(sin(n*Pi*x/L)*(a[n]*cos(n*Pi*c*t/L)

+b[n]*sin(n*Pi*c*t/L)),n=1..N);

ψ := N →
N∑

n=1

sin(
nπ x

L
) (an cos(

n π c t

L
) + bn sin(

n π c t

L
))

The given initial spatial (f(x)) and velocity (g(x)) profiles are entered.
> f(x):=0:

g(x):=piecewise(x<L/4,4*v*x/L,x<L/2,(4*v/L)*(L/2-x),x<L,0);

g(x) :=

⎧⎪⎪⎪⎪⎨
⎪⎪⎪⎪⎩

4 v x

L
x <

L

4
4 v (

L

2
− x)

L
x <

L

2
0 x < L

The coefficients an and bn are explicitly evaluated.
> a[n]:=(2/L)*int(f(x)*sin(n*Pi*x/L),x=0..L);

an := 0
> b[n]:=simplify((2/(n*Pi*c))*int(g(x)*sin(n*Pi*x/L),x=0..L));

bn := −
8 v L (sin(

nπ

2
) − 2 sin(

nπ

4
))

n3 π3 c
The first 25 terms in the Fourier series are calculated, being left as formal sum.

> sol:=psi(25);

sol :=
25∑

n=1

⎛
⎜⎝−

8 sin(
nπ x

L
) v L (sin(

nπ

2
) − 2 sin(

nπ

4
)) sin(

n π c t

L
)

n3 π3 c

⎞
⎟⎠

For animation purposes, the parameter values are substituted into the solution.
The transverse velocity of the string is then calculated so that we can decide if
25 terms is sufficient to fit the initial velocity profile and thus ensure an accurate
animation of the string.

> sol2:=subs({L=20,v=5,c=1},sol): vel:=diff(sol2,t):

Substituting the parameter values into g(x) and evaluating the velocity at t=0,
g(x) and vel are now plotted in the same figure, the former curve being colored
blue, the latter colored red. The scaling is constrained. The resulting picture
is reproduced in Figure 4.2.

> plot([subs({L=20,v=5,c=1},g(x)),eval(vel,t=0)],x=0..20,color
=[blue,red],thickness=2,scaling=constrained,tickmarks=[4,4]);

Figure 4.2: Fourier series superimposed on initial velocity profile.

Clearly, 25 terms produces a very good fit to the initial velocity profile. The
motion of the string, described by sol2 , is then animated.

> animate(sol2,x=0..20,t=0..50,frames=100,thickness=2,

scaling=constrained,tickmarks=[4,3]);
What do you think happens? You will have to execute the recipe to find out.

4.1.4 The 3-Piece String

What is called an acute knowledge of human nature is mostly noth-
ing but the observer’s own weaknesses reflected back from others.
G. C. Lichtenberg, German physicist, philosopher, (1742–99)

In this recipe, we will derive the energy reflection and transmission coefficients
for a monochromatic (frequency ω) plane wave traveling from x = −∞ in an
infinitely long string which has a different constant linear density in the region
x=0 to L than in the rest of the string. If the linear density in region 1 (x < 0)
is ε, in region 2 (0 < x < L) is ε2, and in region 3 (x > L) is ε again, the
corresponding wave numbers are k1 = k = ω

√
ε/T , k2 = ω

√
ε2/T = r k with

r ≡√(ε2/ε), and k3 =k, respectively, where T is the tension in the string.
Using complex notation, in region 1 the wave form is ψ1 = ei(kx−ωt) +

b1 e−i(kx+ωt). (The physical solution is the real part of ψ1.) The spatial part
is now entered, e−iωt being omitted since it will cancel.

> restart:

> psi||1:=exp(I*k*x)+b1*exp(-I*k*x);

ψ1 := e(k x I) + b1 e(−I k x)

The first term is the “incident” wave with unit amplitude1 traveling in the
1Since the reflection and transmission coefficients involve ratios, the incident wave ampli-

tude can be chosen to be 1 without any loss of generality.

positive x-direction, the second the “reflected” wave with amplitude b1 traveling
in the negative x-direction. The energy reflection coefficient, which measures
the fraction of the incident wave energy in region 1 which is reflected, then is
R= |b1 |2 =(b1) (b1 ∗), where the star denotes complex conjugate.

The spatial part of the solution in region 2, which is a linear combination of
plane waves traveling in the positive and negative x directions, is entered. The
amplitudes are labeled a2 and b2 .

> psi||2:=a2*exp(I*r*k*x)+b2*exp(-I*r*k*x);

ψ2 := a2 e(r k x I) + b2 e(−I r k x)

In region 3, there will only be a plane wave traveling in the positive x-direction.
> psi||3:=a3*exp(I*k*x);

ψ3 := a3 e(k x I)

The energy transmission coefficient, which measures the fraction of the incident
energy in region 1 which is transmitted into region 3, will be given by T = |a3 |2.
Clearly, since energy must be conserved, one must have R+T =1. To determine
R and T , the four unknown coefficients must be determined. The string is
continuous, so ψ1 = ψ2 at x = 0 and ψ2 = ψ3 at x = L. These boundary
conditions are entered in eq1 and eq2 .

> eq||1:=eval(psi||1=psi||2,x=0);

eq1 := 1 + b1 = a2 + b2
> eq||2:=eval(psi||2=psi||3,x=L);

eq2 := a2 e(r k L I) + b2 e(−I r k L) = a3 e(k L I)

The continuity of the slopes at x=0 and L is entered in eq3 and eq4 .
> eq||3:=eval(diff(psi||1=psi||2,x),x=0);

eq3 := k I − b1 k I = a2 r k I − b2 r k I

> eq||4:=eval(diff(psi||2=psi||3,x),x=L);

eq4 := a2 r k e(r k L I) I − b2 r k e(−I r k L) I = a3 k e(k L I) I

The sequence of four equations is solved for b1 , a2 , b2 , and a3 .
> sol:=solve({seq(eq||i,i=1..4)},{b1,a2,b2,a3}); assign(sol):

On assigning the solution, R is calculated by multiplying b1 by its complex
conjugate, employing the complex evaluation command, and simplifying.

> R:=simplify(evalc(b||1*conjugate(b||1)));

R :=
(−1 + cos(r k L)2) (r2 − 1)2

−2 r2 cos(r k L)2 − r4 + r4 cos(r k L)2 − 2 r2 − 1 + cos(r k L)2

The energy transmission coefficient T =(a3 ) (a3 ∗) is similarly calculated.
> T:=simplify(evalc(a||3*conjugate(a||3)));

T := − 4 r2

−1 − 2 r2 + cos(r k L)2 − 2 r2 cos(r k L)2 + r4 cos(r k L)2 − r4

As a check, let’s confirm that the sum S ≡ R + T =1.

> S:=simplify(R+T);

S := 1
To get a feeling for the behavior of R and T , let’s choose, say, r=2, i.e., region
2 has a density 4 times that of regions 1 and 3. Setting k L=z in R and T , the
reflection and energy coefficients are given by T1 and R1 ,

> r:=2: T1:=algsubs(k*L=z,T); R1:=algsubs(k*L=z,R);

T1 := − 16
−25 + 9 cos(2 z)2

R1 :=
−9 + 9 cos(2 z)2

−25 + 9 cos(2 z)2
which are then plotted along with S =1 over the range z=0 to 6.

> plot([R1,T1,S],z=0..6,color=[red,blue,green],thickness=2,

labels=["kL","R,T"],numpoints=100);

Figure 4.3: R and T for the 3-piece string as a function of k L.

The bottom oscillatory curve is R, the top oscillatory curve is T , and the
horizontal line is the sum S of the two. Whenever r z ≡ r k L = nπ, with
n = 0, 1, 2, ..., the reflection coefficient R is 0 and the transmission coefficient
T is 1 (100% transmission).

4.1.5 Encore?

Applause is a receipt, not a bill.
Artur Schnabel, American pianist, on why he didn’t give encores, (1882–1951)

Early in my academic career, I was assigned the task of teaching the fresh-
man physics class of several hundred in a huge tiered lecture hall. Being a
theoretician, I approached this task with some trepidation, knowing that in
order to demonstrate some basic ideas of physics, and to keep the class awake,

I would have to intersperse the lecture material with demonstrations. In one
demo, a heavy iron ball was suspended from a long vertical rope attached to a
hidden catwalk far above the lecture podium. Standing with my back against
a 4 × 8 plywood sheet, I would pull the heavy ball away from the vertical a
sizeable distance, bring it up to my chin, and release it. I would stand there
bravely, lecturing on the conservation of energy, as the ball completed a few
oscillations without, of course, hitting my chin.

This seemed a bit tame, so I decided to add a humorous twist. A depart-
mental assistant was placed out of sight on the catwalk with instructions to give
the rope a good heave on the third swing. I would step away from the plywood
sheet after the second swing and the iron ball would crash into the plywood.
I would then express my relief at having avoided injury by a clear violation of
energy conservation. Unfortunately, either the assistant couldn’t count or was
paid off by the students. He pushed on the second swing! As the ball hurtled
towards my chin at an alarming speed, I knew that something was wrong, but
I reacted slowly. I got my hands up in time to avoid serious injury, but the
momentum of the ball knocked me against the plywood sheet which toppled
with a crash to the floor. The students loved it, whistling and cheering and
demanding an encore. Somewhat dazed, I declined!

However, I did repeat the demo the next year with a better trained assis-
tant, replacing the plywood with a large plate of old glass painted black so
the students didn’t know it was glass. This time it was a dazzling success as I
stepped away in time and the ball shattered the glass. The down side was that
I had to sweep up all the glass, as the janitorial staff refused to do so.

Although the following recipe is not identical to the situation described
above, it is inspired by that early exciting demo. A small iron ball of mass
M is attached to the lower end of a long vertical rope of length L which has
a uniform linear density ε. Derive the equation of motion for small transverse
oscillations of the rope. Solve the equation of motion for the normal modes of
oscillation if the initially vertical rope is given a non-zero transverse velocity.
Taking M = 10 kg, L = 10 m, ε = 0.1 kg/m, and the gravitational acceleration
g=10 m/s2, animate the normal mode with the lowest frequency.

The plots library package is needed for the animation and the plottools
package required in order to rotate the animation by 90◦.

> restart: with(plots): with(plottools):

Taking the origin at the top of the rope and measuring the vertical distance y
downwards, the tension T in the rope will be given by T = M g + ε (L − y) g.
At the bottom of the rope (y=L), the upward tension has to only balance the
weight of the ball, but at the top of the rope (y=0) it has to balance both the
weight of the ball and the weight of the entire rope.

> T:=M*g+epsilon*(L-y)*g;

T := M g + ε (L − y) g

The transverse (horizontal) oscillations of the rope will be described by the 1-
dimensional wave equation ∂(T ∂ψ/∂y)/∂y=ε ∂2ψ/∂t2. On entering this PDE,

the tension is automatically substituted, yielding the equation of motion pde.
> pde:=diff(T*diff(psi(y,t),y),y)=epsilon*diff(psi(y,t),t,t);

pde := −ε g (
∂

∂y
ψ(y, t)) + (M g + ε (L − y) g) (

∂2

∂y2 ψ(y, t)) = ε (
∂2

∂t2
ψ(y, t))

Since the rope initially has zero displacement, but a non-zero velocity, a normal-
mode solution of the form ψ(y, t)=X(y) sin(ω t) is sought. The pdsolve com-
mand is applied, with the assumptions that ω, ε, M , L, and g are positive, and
y < L.

> sol:=pdsolve(pde,psi(y,t),HINT=X(y)*sin(omega*t),INTEGRATE,

build) assuming omega::positive, epsilon::positive,
M::positive,L::positive, g::positive, y<L;

sol := ψ(y, t) = sin(ω t) C1 BesselJ(0,
2 ω

√
M + εL − ε y√

ε
√

g
)

+ sin(ω t) C2 BesselY(0,
2 ω

√
M + εL − ε y√

ε
√

g
)

The spatial part of the solution is a linear combination of zeroth-order Bessel
functions of the first and second kinds. The spatial part X is now extracted.

> X:=simplify(rhs(sol)/sin(omega*t),symbolic);

X:= C1 BesselJ(0,
2 ω

√
M + εL − ε y√

ε
√

g
)+ C2 BesselY(0,

2 ω
√

M + εL − ε y√
ε
√

g
)

Assuming that the free-end boundary condition (dX/dy)|y=L = 0 applies, we
solve for C1 .

> _C1:=solve(subs(y=L,diff(X,y))=0,_C1);

C1 := −
C2 BesselY(1,

2 ω
√

M√
ε
√

g
)

BesselJ(1,
2 ω

√
M√

ε
√

g
)

The parameter values are entered and X divided by the arbitrary constant C2
and expanded to yield X2 .

> L:=10: M:=10: g:=10: epsilon:=1/10: X2:=expand(X/_C2);

X2 := −
BesselY(1, 2 ω

√
10) BesselJ(0, 2 ω

√
− y

10
+ 11)

BesselJ(1, 2 ω
√

10)

+ BesselY(0, 2 ω

√
− y

10
+ 11)

The fixed-end boundary condition at y=0 is applied to the numerator of X2 .
> bc:=eval(numer(X2),y=0)=0;

bc := −BesselY(1, 2 ω
√

10) BesselJ(0, 2 ω
√

11)

+ BesselY(0, 2 ω
√

11) BesselJ(1, 2 ω
√

10) = 0
The left-hand side of bc is plotted in Figure 4.4 over the frequency range ω=0.1
to 20 and the vertical view limited so as to be able to clearly see the zeros.

> plot(lhs(bc),omega=0.1..20,view=[0..20,-0.2..0.1]);

Figure 4.4: Plot of lhs of the boundary condition at y=0 versus frequency.

Guided by the plot, the lowest zero is numerically sought between ω=4 and 6.
> omega:=fsolve(bc,omega=4..6);

ω := 5.137764180
The lowest possible frequency for a normal mode is ω = 5.14 Hz. The mode
with the lowest frequency then is given by ψ. Higher frequency normal modes
can be obtained by selecting other zeros in the fixed-end boundary condition.

> psi:=X2*sin(omega*t);

ψ :=

(
−

BesselY(1, 10.27552836
√

10) BesselJ(0, 10.27552836
√

− y

10
+ 11)

BesselJ(1, 10.27552836
√

10)

+ BesselY(0, 10.27552836
√

− y

10
+ 11)

)
sin(5.137764180 t)

The solution ψ is animated, the scaling being left unconstrained for better
viewing. Using the rotate command, the figure is rotated by −90◦ so that the
string is initially vertical and the subsequent time-dependent displacement is
horizontal.

> rotate(animate(psi,y=0..L,t=0..5,color=red,

thickness=2,frames=50,numpoints=100),-Pi/2);
You will have to execute the worksheet on your computer to see the animation.

4.2 Beyond the String

Many more string examples are given in the Supplementary Recipes, but
it’s time to move on to other physical examples and also look at two- and
three-dimensional situations. All the recipes in this section involve Cartesian
coordinates. The next section is devoted entirely to non-Cartesian examples.

4.2.1 Heaviside’s Telegraph Equation

Get your facts first,
and then you can distort them as much as you please.
Mark Twain, American author on propaganda, (1835–1910)

Consider an electrical transmission line carrying a current I(x, t) that has a
uniform inductance L, capacitance C to ground (zero potential), and resistance
R (all per unit length) and has a leakage (leakage coefficient G per unit length)
proportional to the potential V from the line to ground.

(a) Show that V (x, t) satisfies the telegraph equation,
∂2V

∂x2 = L C
∂2V

∂t2
+ (R C + L G)

∂V

∂t
+ R GV. (4.6)

(b) Under what conditions does the telegraph equation reduce to a wave equa-
tion? a diffusion equation? Identify the wave velocity for the former, the
diffusion coefficient for the latter.

(c) Show that a general solution of the form V (x, t)=e−k tf(α x+β t), where
f is an arbitrary function, will satisfy (4.6), provided that R C =L G and
k, α, and β take on certain forms which are to be determined. Briefly
discuss what this solution means.

This problem was first solved by the English electrical engineer Oliver Heaviside
in 1887. Heaviside also reduced Maxwell’s equations into the simpler form
that we now know, invented the differential operator (D) notation for solving
differential equations, predicted the existence of an ionized reflective layer (the
Heaviside layer) in the ionosphere which bounces radio signals back to earth,
and predicted that the mass of an electric charge would increase with velocity.

Now let’s answer the question. In order to use the symbol I for the current,
the imaginary unit

√−1 is set equal to j with the following interface command.
> restart: interface(imaginaryunit=j):

The rate of change of the voltage V with distance x is given by eq1 . The first
term −R I on the rhs is the potential drop per unit length due to the resistance,
which is given by Ohm’s law. The second term −L ∂I/∂t is the back emf per
unit length due to the inductance.

> eq1:=diff(V(x,t),x)=-R*I(x,t)-L*diff(I(x,t),t);

eq1 :=
∂

∂x
V (x, t) = −R I (x, t) − L (

∂

∂t
I (x, t))

The rate of change of the current I with x is given by eq2 . The first term on
the rhs is the leakage of current per unit length from the line to ground, the
second term the capacitive loss per unit length to ground.

> eq2:=diff(I(x,t),x)=-G*V(x,t)-C*diff(V(x,t),t);

eq2 :=
∂

∂x
I (x, t) = −GV (x, t) − C (

∂

∂t
V (x, t))

eq1 and eq2 are differentiated with respect to x, t, respectively, in eq3 and eq4 .

> eq3:=diff(eq1,x); eq4:=diff(eq2,t);

eq3 :=
∂2

∂x2V (x, t) = −R (
∂

∂x
I (x, t)) − L (

∂2

∂x ∂t
I (x, t))

eq4 :=
∂2

∂x ∂t
I (x, t) = −G (

∂

∂t
V (x, t)) − C (

∂2

∂t2
V (x, t))

A PDE for V (x, t) alone is obtained by substituting eq2 and eq4 into eq3 .
Collecting the ∂V/∂t terms then yields the telegraph equation TE .

> eq5:=subs({eq2,eq4},eq3);
> TE:=collect(eq5,diff(V(x,t),t));

TE:=
∂2

∂x2V (x, t) = (R C + L G) (
∂

∂t
V (x, t)) + R GV (x, t) + L C (

∂2

∂t2
V (x, t))

The wave equation WE follows on setting R = 0 and G = 0 in TE , the wave
velocity being c=1/

√
L C.

> WE:=eval(TE,{R=0,G=0});

WE :=
∂2

∂x2V (x, t) = L C (
∂2

∂t2
V (x, t))

The diffusion equation DE results on setting L = 0 and G = 0 in TE , the
diffusion coefficient being d= 1/(R C).

> DE:=eval(TE,{L=0,G=0});

DE :=
∂2

∂x2V (x, t) = R C (
∂

∂t
V (x, t))

To answer the last part of the question, the given ansatz is entered.

> ansatz:=V(x,t)=exp(-k*t)*f(alpha*x+beta*t);

ansatz := V (x, t) = e(−k t) f (α x + β t)
The pdetest command is used to test whether the proposed solution sat-
isfies the telegraph equation TE . If it does, the answer would be zero, but
instead it yields an algebraic equation, appropriately in terms of differential
operators, the notation introduced by Heaviside.

> eq6:=pdetest(ansatz,TE);

eq6 := −e(−k t)(−(D(2))(f)(α x + β t) α2 − R C k f (α x + β t)
+ R C D(f)(α x + β t) β − L G k f (α x + β t) + L G D(f)(α x + β t) β

+ R G f (α x + β t) + L C k2 f (α x + β t) − 2 L C k D(f)(α x + β t) β

+ L C (D(2))(f)(α x + β t) β2)

By inspection of the output in eq6 , the second-order differential terms will
cancel if we choose β = α/

√
L C. Assuming that G = R C/L, all remaining

terms will cancel provided that k = R/L. To confirm that this is so, these
relations are entered and the new ansatz, ansatz2 displayed, and entered as the
argument in pdetest.

> beta:=alpha/sqrt(L*C): G:=R*C/L: k:=R/L: ansatz2:=ansatz;

ansatz2 := V (x, t) = e(−R t
L ) f (α x +

α t√
L C

)

> pdetest(ansatz2,TE);

0
The answer is zero, indicating that a solution to the telegraph equation exists
for G=R C/L of the form V =e−R t/Lf(α(x + t/

√
L C)). This solution implies

“distortionless” propagation of the input wave form with velocity 1/
√

L C, the
wave form decreasing in amplitude with time.

4.2.2 Spiegel’s Diffusion Problems

Any solution to a problem changes the problem.
R. W. Johnson, American journalist, Washingtonian, Nov. 1979

Murray Spiegel’s Advanced Mathematics [Spi71] is probably the classic source
book of solved problems relevant to mathematical physics. The following recipe
is based on two diffusion examples (Problems 12.16 and 12.17) taken from that
text. Of course, we shall use computer algebra here, instead of performing a
hand calculation. I shall take each example one step further by animating the
solutions. Simply staring at the series solutions is no substitute for observing
what actually happens as time progresses. So here are the problems.

(a) Using the method of separation of variables, solve the 1-dimensional heat
diffusion equation in a thin bar 3 m in length whose surface is insu-
lated and has a diffusion coefficient of 2 m2/s. Its ends are kept at
0 ◦C (T (0, t) = T (3, t) = 0) and it has the initial temperature profile
T (x, 0)=5 sin(4 π x) − 3 sin(8 π x) + 2 sin(10 π x). Animate the tempera-
ture profile over the time interval t=0 to 0.025 seconds.

(b) If the bar in part (a) has an initial internal constant temperature of 25 ◦C
with the ends held at 0 ◦C, use the Fourier series method to derive the
temperature distribution for arbitrary time t > 0. Animate the solution
over the time interval t=0 to 2 seconds, keeping 20 terms in the series.

After loading the plot package, needed for the animations, the 1-dimensional
heat diffusion equation for the bar of length L=3 and diffusion coefficient d=2
is entered in pde.

> restart: with(plots): d:=2: L:=3:

> pde:=diff(T(x,t),t)=d*diff(T(x,t),x,x);

pde :=
∂

∂t
T (x, t) = 2 (

∂2

∂x2 T (x, t))

The pdsolve command is applied to pde. Since the separation of variables
method is requested, the hint that T (x, t) = X(x) Y (t) is provided. The sepa-
rated ODEs are integrated and the product solution built in sol .

> sol:=pdsolve(pde,HINT=X(x)*Y(t),INTEGRATE,build);

sol := T (x, t) = C3 (e( c1 t))2 C1 e(
√

c1 x) +
C3 (e( c1 t))2 C2

e(
√

c1 x)

Without loss of generality the integration constant C3 can be set equal to 1
on the rhs of sol . For later convenience, C1 and C2 are relabeled as A and
B, respectively, and the separation constant c1 =−k2.

> T:=subs({_C3=1,_C1=A,_C2=B,_c[1]=-kˆ2},rhs(sol));

T := (e(−k2 t))2 A e(
√−k2 x) +

(e(−k2 t))2 B

e(
√−k2 x)

T is simplified with the symbolic option in T2 .
> T2:=simplify(T,symbolic);

T2 := A e(−k (2 k t−x I)) + B e(−k (2 k t+x I))

To satisfy the boundary condition T (0, t) = 0, we must set B = −A in T2 .
The complex evaluation command is then applied so as to convert the complex
exponential into a sine function.

> T3:=evalc(subs(B=-A,T2));

T3 := 2 I A e(−2 k2 t) sin(k x)
To satisfy the boundary condition T (3, t)=0, we must have sin(3 k)=0, so k=
m π/3, with m=1, 2, 3, .... The coefficient is removed from T3 by substituting
A=1/(2 I).

> k:=m*Pi/L: T4:=subs(A=1/(2*I),T3);

T4 := e(− 2 m2 π2 t
9 ) sin(

m π x

3
)

The general solution is of the form T (x, t)=
∑∞

m=1 Cm e(− 2 m2 π2 t
9 ) sin(

m π x

3
).

The initial temperature profile will be satisfied by retaining the three terms
corresponding to m = 12, 24, and 30, with C12 = 5, C24 = −3, and C30 = 2.
Using T4 , this is done in T5 .

> T5:=5*eval(T4,m=12)-3*eval(T4,m=24)+2*eval(T4,m=30);

T5 := 5 e(−32 π2 t) sin(4π x) − 3 e(−128 π2 t) sin(8π x) + 2 e(−200 π2 t) sin(10 π x)

The temperature profile given in T5 is animated over the time interval t=0 to
0.025 s, the opening frame of the animation (initial temperature profile) being
shown in Figure 4.5. As t increases, T (x, t) decays to zero inside the bar.

> animate(T5,x=0..L,t=0..0.025,frames=100,numpoints=500,

thickness=2,labels=["x","T"]);

–8

0

T

8

1 2 3
x

Figure 4.5: The initial temperature profile in the bar for part (a).

Now let’s tackle part (b) of the problem. The initial temperature profile is
a constant 25 ◦C everywhere inside the bar. Using the Fourier series approach,
the general Fourier coefficient is given by Cm = (2/L)

∫ L

0 25 sin(m π x/L) dx
with L = 3 and m a positive integer. This coefficient is now calculated.

> C[m]:=(2/L)*int(25*sin(m*Pi*x/L),x=0..L) assuming m::integer;

Cm := −50 (−1 + (−1)m)
m π

The formal Fourier series representation (out to m=20) of the temperature is
displayed in Temp, and then explicitly calculated with the value command.

> Temp:=Sum(C[m]*T4,m=1..20);

Temp :=
20∑

m=1

⎛
⎜⎝−

50 (−1 + (−1)m) e(− 2 m2 π2 t
9 ) sin(

m π x

3
)

m π

⎞
⎟⎠

> Temp:=value(Temp);

Temp :=
100 e(− 2 π2t

9 ) sin(
πx

3
)

π
+

100
3

e(−2 π2t) sin(πx)
π

+
20 e(− 50 π2t

9 ) sin(
5πx

3
)

π
+

100
7

e(− 98 π2t
9 ) sin(

7πx

3
)

π
+

100
9

e(−18 π2 t) sin(3π x)
π

+
100
11

e(− 242 π2t
9 )sin(

11πx

3
)

π
+

100
13

e(− 338 π2t
9 )sin(

13πx

3
)

π
+

20
3

e(−50 π2t)sin(5πx)
π

+
100
17

e(− 578 π2t
9 ) sin(

17πx

3
)

π
+

100
19

e(− 722 π2t
9 ) sin(

19πx

3
)

π

The temperature distribution is now animated over the time interval t=0 to 2
seconds, the initial frame in the animation being shown in Figure 4.6.

> animate(Temp,x=0..L,t=0..2,frames=50,thickness=2,

numpoints=500,labels=["x","T"]);

Figure 4.6: The initial temperature profile for part (b) with 20 terms.

Because of the step function nature of the initial temperature profile at the
ends of the bar, Gibb’s oscillations appear in the initial series representation,
oscillating around the exact profile. The oscillations quickly disappear as the
animation progresses and once again T (x, t) decreases to zero inside the bar.

4.2.3 Introducing Laplace’s Equation

An editor is someone who separates the wheat from the chaff
and then prints the chaff.
Adlai Stevenson, American politician, referring to news reporting (1900–65)

In electrostatics, the electric field �E is related to the electric potential V by
�E = −∇V . In SI units, Maxwell’s equations are ∇ × �E = 0 and ∇ · �E = ρ/ε0,
where ρ is the electric charge density and ε0 the permittivity of free space. Sub-
stituting �E into the first relation yields the vector identity ∇ × ∇V =0, while
the second relation produces Poisson’s equation, ∇2V =−ρ/ε0. In charge-free
regions of space, ρ = 0 and Poisson’s equation reduces to Laplace’s equation
∇2V =0. This recipe involves solving a two-dimensional electrostatic boundary-
value problem, using Laplace’s equation in Cartesian coordinates.

Determine the potential V (x, y) in the charge-free region x ≥ 0, 0 ≤ y ≤ a,
given that the boundary at x=0 is held at the constant potential V0, while the
boundaries at y=0 and a are held at zero potential. The asymptotic boundary
condition is V → 0 for x → ∞. Taking a= 1 meter and V0 = 2 volts, produce
two- and three-dimensional contour plots of V (x, y) for x ≥ 0, 0 ≤ y ≤ a.

After loading the plots library package, needed for the contour plots, the
two-dimensional form of Laplace’s equation is entered for the potential V (x, y).

> restart: with(plots):

> pde:=diff(V(x,y),x,x)+diff(V(x,y),y,y)=0;

pde := (
∂2

∂x2 V (x, y)) + (
∂2

∂y2 V (x, y)) = 0

pde is analytically solved, assuming that V (x, y)=X(x) Y (y).
> sol:=pdsolve(pde,HINT=X(x)*Y(y),INTEGRATE,build);

sol := V (x, y) = C3 sin(
√

c1 y) C1 e(
√

c1 x) +
C3 sin(

√
c1 y) C2

e(
√

c1 x)

+ C4 cos(
√

c1 y) C1 e(
√

c1 x) +
C4 cos(

√
c1 y) C2

e(
√

c1 x)

The substitution c1 =k is made on the rhs of sol and the result simplified.
> sol:=simplify(subs(sqrt(_c[1])=k,rhs(sol)));

sol := ( C3 sin(k y) C1 e(2 k x) + C3 sin(k y) C2

+ C4 cos(k y) C1 e(2 k x) + C4 cos(k y) C2 ) e(−k x)

To satisfy the boundary condition V =0 at y=0, the cos(k y) term is removed
from sol by setting it equal to zero. To satisfy V = 0 at y = a for arbitrary
x > 0, we must have sin(k a) = 0, so k = nπ/a, where n is a positive integer.
The term e2 k x must also be removed so that V → 0 as x → ∞.

> sol2:=subs({cos(k*y)=0,exp(2*k*x)=0,k=n*Pi/a},sol);
sol2 := C3 sin(

nπ y

a
) C2 e(−n π x

a )

Using the operand command, op, the coefficient combination is replaced with
A. The arguments may have to be changed if the terms are ordered differently.

> sol3:=A*op(2,sol2)*op(4,sol2);

sol3 := A sin(
nπ y

a
) e(−n π x

a )

Making use of orthogonality of the sine functions over the interval y = 0 to a,
the coefficient A is calculated, assuming that n is an integer.

> A:=(2/a)*int(V0*sin(n*Pi*y/a),y=0..a) assuming n::integer;

A := −2V0 (−1 + (−1)n)
nπ

Entering the given values of the parameters, the result in sol3 is summed. A
large number of terms (100) is kept to obtain smooth contour plots.

> a:=1: V0:=2: V:=sum(sol3,n=1..100):

Using the contourplot command, V is plotted over the interval x = 0 to 2
and y = 0 to a = 2. Constant potential curves (equipotentials) are plotted for
0.2, 0.4,...,1.6, 1.8 volts. The grid is taken to be 50 × 50 in order to obtain
reasonably smooth curves. The resulting picture is shown in Figure 4.7, the 0.2
volt curve being furthest to the right.

> contourplot(V,x=0..2,y=0..a,contours=[seq(i*V0/10,i=1..9)],

thickness=2,grid=[50,50]);

0

0.2

0.4

y

0.8

1

0.2 0.4 x 0.8

Figure 4.7: Equipotentials corresponding to 0.2 (far right), 0.4,...,1.8 volts.

The contourplot3d command is used to produce a three-dimensional con-
tour plot, the equipotential curves again being spaced 0.2 volts apart.

> contourplot3d(V,x=0..1,y=0..a,contours=[seq(i*V0/10,i=1..9)],

grid=[60,60],shading=zhue,filled=true,axes=boxed,
tickmarks=[2,2,2]);

The resulting picture can be viewed by executing the recipe on your computer.

4.2.4 Grandpa’s “Trampoline”

In the first place God made idiots. This was for practice.
Then He made School Boards.
Mark Twain, Following the Equator, (1897)

From his sundeck, my grandson Daniel can peek over the fence and see his
neighbor’s children happily bouncing up and down on their trampoline after
coming home from school. He is too young to join in the fun, so must be con-
tent (not very!) to watch. On the other hand, I am getting too old to engage in
trampoline antics, being more content to substitute the following recipe instead.

A trampoline, consisting of a square horizontal elastic membrane fixed on
its four edges (x=0, b and y=0, b), has an initial transverse profile ψ(x, y, 0) ≡
f = A x2 y (b − x) (b − y)3 and is released from rest. Determine ψ(x, y, t) for
arbitrary time t > 0. Taking A = 1/5 m−7, b = 2 m, and the wave speed c = 1
m/s, animate the trampoline motion.

The plots package is loaded, because we shall be animating the motion of the
trampoline. A functional operator F is created to calculate the second derivative
of ψ(x, y, t) with respect to an arbitrary variable v.

> restart: with(plots): F:=v->diff(psi(x,y,t),v,v):

The transverse vibrations of the trampoline are governed by the two-dimensional
wave equation which is entered using F.

> pde:=F(x)+F(y)=F(t)/cˆ2;

pde := (
∂2

∂x2 ψ(x, y, t)) + (
∂2

∂y2 ψ(x, y, t)) =

∂2

∂t2
ψ(x, y, t)

c2

The wave equation is solved by the method of separation of variables, the as-
sumed form ψ(x, y, t)=X(x) Y (y) T (t) being supplied as a hint.

> sol:=pdsolve(pde,HINT=X(x)*Y(y)*T(t),INTEGRATE,build);

The output (not displayed here) has two separation constants, c1 and c2. To
simplify the form of the solution, the substitution c1 = −α2 and c2 = −β2 is
made on the rhs of sol and the result simplified with the symbolic option.

> psi1:=simplify(subs({_c[1]=-alphaˆ2,_c[2]=-betaˆ2},
rhs(sol)),symbolic);

ψ1 := C5 sin(c
√

α2 + β2 t) C3 C1 e((β y+α x) I) + · · ·
+ C6 cos(c

√
α2 + β2 t) C3 C1 e((β y+α x) I) + · · ·

The initial transverse velocity is zero, so the time-dependent sine terms must
be removed from ψ1.

> psi2:=remove(has,psi1,sin);

The exponential terms in ψ2 are converted to a trig form in ψ3 and the result
expanded. The terms cos(β y), sin(β y), cos(α x), and sin(α x) appear.

> psi3:=expand(convert(psi2,trig));

To satisfy the fixed edge conditions along y=0 and x=0, the cosine terms are
removed from the solution by setting them equal to zero, while one must have
α=m π/b and β=nπ/b , with m, n=1, 2, 3, ..., to satisfy the conditions along
x=b and y=b. Making these substitutions and factoring yields ψ4.

> psi4:=factor(subs({cos(beta*y)=0,cos(alpha*x)=0,
alpha=m*Pi/b,beta=n*Pi/b},psi3));

ψ4:= −cos(c π

√
m2 + n2

b2 t) sin(
nπ y

b
) sin(

m π x

b
) C6 ( C1 − C2 )( C3 − C4 )

The following select command is used to replace the cumbersome coefficient

combination in ψ4 with the symbol Bm,n.
> psi5:=B[m,n]*select(has,psi4,{sin,cos});

ψ5 := Bm,n cos(c π

√
m2 + n2

b2 t) sin(
nπ y

b
) sin(

m π x

b
)

The initial transverse profile of the trampoline is entered.
> f:=A*xˆ2*y*(b-x)*(b-y)ˆ3;

f := A x2 y (b − x) (b − y)3

Using orthogonality of the sine functions, the coefficient Bm,n is evaluated by
performing the double integration

Bm,n =(2/b)2
∫ b

0

∫ b

0
f sin(m π x/b) sin(nπ y/b) dx dy,

and assuming that m and n are integers.
> B[m,n]:=(2/b)ˆ2*int(int(f*sin(m*Pi*x/b)*sin(n*Pi*y/b),

x=0..b),y=0..b) assuming m::integer,n::integer;

Bm, n := −48 b7 A(−4 + 8 (−1)(1+m) + n2 π2 + 2 (−1)m π2 n2 + 4 (−1)n

+8 (−1)(n+m))
/

(m3 π8 n5)

The solution then will be of the form

ψ(x, y, t) =
∞∑

m=1

∞∑
n=1

Bm,n sin(m π x/b) sin(nπ y/b) cos(c π
√

m2 + n2 t/b).

A functional operator G is formed, using ψ5, for explicitly calculating this double
Fourier series out to N terms in each sum,

> G:=N->sum(sum(psi5,m=1..N),n=1..N):

The given parameters are entered, and the double series calculated for N =10.
You will have to execute the recipe on your own computer to see the 10 × 10
= 100 terms in ψ.

> A:=1/5: b:=2: c:=1: N:=10: psi:=G(N);

Using the animate command, the answer ψ is animated over the time interval
t=0 to 5 seconds, 50 frames being used. Execute the recipe and enjoy!

> animate(plot3d,[psi,x=0..b,y=0..b],t=0..5,frames=50,

axes=boxed,shading=zhue,tickmarks=[3,3,3]);

4.2.5 Irma Insect’s Isotherm

A man thinks he amounts to a great deal but to a flea or a mosquito
a human being is merely something good to eat.
Don Marquis, American humorist, (1878–1937)

In the Erehwon zoo, Irma insect lives inside a rectangular enclosure with walls
at x = 0, a and y = 0, b, floor at z = 0 and ceiling at z = c. In the winter, the

floor and walls have a temperature of zero degrees while the ceiling has a steady
temperature profile T (x, y, c)=1600x (a − x) y (b − y) degrees. Determine the
temperature inside Irma’s enclosure. Irma is most comfortable when the tem-
perature is 20 ◦. If a = b = c = 1, plot the 3-dimensional isothermal surface
corresponding to this temperature. What is T at the center of the enclosure?

The steady-state temperature inside the enclosure is given by the solution of
Laplace’s equation, ∇2T (x, y, z)=0. For variety, let’s load the VectorCalculus
package and use the Laplacian command in Cartesian coordinates.

> restart: with(plots): with(VectorCalculus):

> pde:=Laplacian(T(x,y,z),’cartesian’[x,y,z])=0;

pde := (
∂2

∂x2 T (x, y, z)) + (
∂2

∂y2 T (x, y, z)) + (
∂2

∂z2 T (x, y, z)) = 0

The temperature of the walls at x = 0 and a is held at zero degrees. On
separating variables, the x part of the solution will be a linear combination
of a sine and a cosine. To satisfy T = 0 at x = 0 and a, this part of the
solution must involve sin(m π x/a), with m a positive integer. Similarly, the y
part of the solution must be sin(nπ y/b), with n a positive integer. So, let’s
apply the pdsolve command with the hint that the solution is of the form
sin(m π x/a) sin(nπ y/b) Z(z), with the structure of Z(z) to be determined.

> sol:=pdsolve(pde,T(x,y,z),HINT=sin(m*Pi*x/a)*

sin(n*Pi*y/b)*Z(z),INTEGRATE,build);

sol := T (x, y, z) = sin(
m π x

a
) sin(

nπ y

b
) C1 sin(

π
√−m2 b2 − n2 a2 z

a b
)

+ sin(
m π x

a
) sin(

nπ y

b
) C2 cos(

π
√−m2 b2 − n2 a2 z

a b
)

Since the floor at z=0 is held at zero degrees, the cosine term must be removed
from the rhs of the solution sol . The result is the general term Tm, n in a double
Fourier series, with the coefficient C1 to be determined.

> T[m,n]:=remove(has,rhs(sol),cos);

Tm, n := sin(
m π x

a
) sin(

nπ y

b
) C1 sin(

π
√−m2 b2 − n2 a2 z

a b
)

The temperature profile at z=c is entered in f . To determine the form of C1 ,
the combination sin(m π x/a) sin(nπ y/b) is first entered in g.

> f:=1600*x*(a-x)*y*(b-y): g:=sin(m*Pi*x/a)*sin(n*Pi*y/b):

Making use of orthogonality of the sine functions, f is multiplied by g and the
double integral over x and y carried out. This must be equal to Tm, n, evaluated
at z=c, multiplied by g, with the same double integration performed.

> eq:=int(int(f*g,x=0..a),y=0..b)

=int(int(eval(T[m,n],z=c)*g,x=0..a),y=0..b):
Then eq is solved for C1, and simplified assuming that m and n are integers.

> _C1:=simplify(solve(eq,_C1)) assuming m::integer,n::integer;

C1 :=
25600 a2 b2 (1 − (−1)m + (−1)m (−1)n − (−1)n)

sin(
π

√−m2 b2 − n2 a2 c

a b
) m3 π6 n3

The dimensions of the enclosure are entered, and the series representation of
the temperature T calculated, keeping 50× 50 = 2500 terms to ensure accurate
numerical results.

> a:=1.0: b:=1.0: c:=1.0: T:=sum(sum(T[m,n],m=1..50),n=1..50):

Setting T = 20, the implicitplot3d command is used to plot the 20 degree
isotherm, the result being the bowl-shaped surface shown in Figure 4.8.

> implicitplot3d(T=20,x=0..a,y=0..b,z=0..c,axes=box,

orientation=[10,70],style=PATCHCONTOUR,tickmarks=[2,2,2]);

Figure 4.8: The 20 ◦C isotherm.

The temperature is now evaluated at the center of the enclosure,
> T2:=evalf(eval(T,{x=a/2,y=b/2,z=c/2}));

T2 := 11.36308962
and found to be about 11 degrees.

4.2.6 Daniel Hits Middle C

The notes I handle no better than many pianists. But the pauses
between the notes – ah, that is where the art resides.
Artur Schnabel, American pianist, Chicago Daily News, 11 June 1958

A horizontal bar of nickel-iron (Young’s modulus, Y = 2.1 × 1011 N/m2, and
density, ρ=7.8×103 kg/m3) of length L=1 m, whose cross-section is rectangu-
lar with width W =0.1 m and height H =0.049 m, is clamped (i.e., ψ=ψ ′ =0)
at x = 0 and L. Daniel strikes the bar sharply at the midpoint of one of its

wider sides in such a way that the initial transverse velocity is (approximately)
v(x, 0) = v0 δ(x − L/2), with v0 = 1 m2/s and δ the Dirac delta function. De-
termine the transverse displacement ψ(x, t) of the bar at arbitrary time t > 0.
Show that the fundamental frequency corresponds to approximately middle C
(f=261.63 Hz). Animate the motion of the bar over the interval t=0 to 10T ,
where T = 1/f is the fundamental period.

After loading the plots library package, needed for the animation,
> restart: with(plots):

the equation of motion for transverse oscillations of the bar is entered.
> pde:=aˆ4*diff(psi(x,t),x$4)+diff(psi(x,t),t,t)=0;

pde := a4 (
∂4

∂x4 ψ(x, t)) + (
∂2

∂t2
ψ(x, t)) = 0

The parameter a=(S κ2 Y/ε)1/4, where S =WH is the cross-sectional area of
the bar, κ = H/

√
12 is the radius of gyration about the horizontal midplane

through the bar, and ε=ρ S is the linear density.
Using pdsolve, pde is solved with the hint ψ(x, t)=X(x) T (t).
> sol:=pdsolve(pde,HINT=X(x)*T(t),INTEGRATE,build);

Setting the separation constant c1 =k4 in sol , the frequency is ω=a2 k2.
> sol:=simplify(subs(_c[1]=kˆ4,sol),symbolic);

sol := ψ(x, t) = e(−I k x) C5 sin(a2 k2 t) C1 + · · · + C6 cos(a2 k2 t) C4 e(k x)

Since the bar is initially horizontal, i.e., ψ(x, 0) = 0, the cosine term must be
removed from the rhs of sol . The result is then factored.

> psi:=factor(remove(has,rhs(sol),cos));

ψ := C5 sin(a2 k2 t)( C1 e(−I k x) + e(−k x) C2 + C3 e(k x I) + e(k x) C4 )
To apply the boundary conditions at the end of the bar, the spatial part is
extracted with the select command. The time part is also selected.

> X:=select(has,psi,x); T:=select(has,psi,t);

X := C1 e(−I k x) + e(−k x) C2 + C3 e(k x I) + e(k x) C4

T := sin(a2 k2 t)
X is converted to trig form, and the cosine, sine, cosh, and sinh terms collected.

> X:=collect(convert(X,trig),[cos,sin,cosh,sinh]);

X := ( C3 + C1 ) cos(k x) + (− C1 I + C3 I) sin(k x)
+ ( C4 + C2 ) cosh(k x) + ( C4 − C2 ) sinh(k x)

Using the op command to extract the relevant operands, new coefficients A1 ,
etc, are introduced into X in the following line.

> X:=add(A||i*op([i,2],X),i=1..4);

X := A1 cos(k x) + A2 sin(k x) + A3 cosh(k x) + A4 sinh(k x)
The clamped-end boundary conditions are applied at x=0 in bc1 and bc2 and
at x=L in bc3 and bc4 .

> bc1:=eval(X,x=0)=0; bc2:=eval(diff(X,x),x=0)=0;

bc1 := A1 + A3 = 0 bc2 := A2 k + A4 k = 0

> bc3:=eval(X,x=L)=0; bc4:=eval(diff(X,x),x=L)=0;

bc3 := A1 cos(k L) + A2 sin(k L) + A3 cosh(k L) + A4 sinh(k L) = 0
bc4 := −A1 sin(k L) k + A2 cos(k L) k + A3 sinh(k L) k + A4 cosh(k L) k = 0

On attempting to solve the four boundary conditions for the four unknown
coefficients, one of the coefficients will be undetermined and a transcendental
equation for k will result. Let’s choose the undetermined coefficient to be A4 .
Then, bc1 , bc2 , and bc3 are solved for A1 , A2 , and A3 and sol2 assigned.

> sol2:=solve({bc1,bc2,bc3},{A||1,A||2,A||3}); assign(sol2):

The coefficient A4 is temporarily set equal to 1 and the fourth boundary con-
dition divided by k3 and simplified with the symbolic option.

> A||4:=1: bc4:=simplify(bc4/kˆ3,symbolic);

bc4 :=
2 (−1 + cos(k L) cosh(k L))
k2 (cos(k L) − cosh(k L))

= 0

We set k L=K in bc4 and isolate the cos(K) term to the left of the equation.
> eq:=isolate(subs(k*L=K,bc4),cos(K));

eq := cos(K) =
1

cosh(K)
On comparing the transcendental equation eq with the corresponding equation
for the clamped-free end situation in Recipe 01-1-3, we see that they differ by
a minus sign on the right-hand side. As before, eq must be solved numerically
for the allowed K values. For large K, cosh(K) becomes very large and cos K
approaches zero. The allowed K values, therefore, are approximately the zeros
of the cosine function. A functional operator f is introduced to determine the
zeros in the range 3(n − 1) to 3 n, where n will take on the values 1, 2, etc.

> f:=n->fsolve(eq,K,3*(n-1)..3*n):

Then forming f(n), dividing by L, and using the sequence command, the first
four zeros of k are given in sol3 which is assigned.

> sol3:=seq(k||n=f(n)/L,n=1..4); assign(sol3):

sol3 := k1 = 0., k2 =
4.730040745

L
, k3 =

7.853204624
L

, k4 =
10.99560784

L
The first zero, k1 , must be rejected, because for k = 0 the lhs of bc4 is finite.
The parameters L, v0, W , H, Y , and ρ are entered, and the radius of gyration
κ, cross-sectional area S, linear density ε, and the parameter a calculated.

> L:=1: v0:=1: W:=0.1: H:=0.049: Y:=2.1*10ˆ11: rho:=7.8*10ˆ3:

> kappa:=evalf(H/sqrt(12)); S:=W*H; epsilon:=rho*S;

a:=(S*kappâ 2*Y/epsilon)̂ (1/4);

κ := 0.01414508160 S := 0.0049 ε := 38.22000 a := 8.567101289
The coefficient A4 , which had been temporarily set equal to 1, is relabeled

B, and will now be calculated. The initial transverse velocity is of the form
v(x, 0)=v0 δ(x−L/2) =

∑∞
n=1 a2 k2

n Bn Xn, where Xn is the nth spatial mode.
Multiplying v(x, 0) by Xn, integrating x over the range 0 to L, and using the
orthogonality of the spatial modes, yields v0 Xn(x=L/2) = Bn a2 k2

n

∫ L

0 X2
n dx,

which is easily solved for Bn. The coefficient B is evaluated in the following
line for a general k value.

> B:=v0*eval(X,x=L/2)/(aˆ2*kˆ2*int(Xˆ2,x=0..L)):

The product B X T is evaluated at k = kn and the first three terms of the
complete solution ψ added. Only the fundamental (n=2) contribution is shown
here, the remaining terms having rapidly decreasing amplitudes.

> psi:=simplify(add(eval(B*X*T,k=k||n),n=2..4));

ψ := −0.0009671479296 sin(1642.092308 t) cos(4.730040745 x)
+ 0.0009502249823 sin(1642.092308 t) sin(4.730040745 x)
+ 0.0009671479296 sin(1642.092308 t) cosh(4.730040745 x)
− 0.0009502249823 sin(1642.092308 t) sinh(4.730040745 x) + · · ·

The fundamental frequency ω=a2 k2 2 is 1642 rads/s or f =ω/(2 π) ≈ 261.3 Hz,
which is close to middle C. The fundamental period T = 1/f is also calculated.

> omega:=aˆ2*k||2ˆ2; f:=evalf(omega/(2*Pi)); T:=1/f;

ω := 1642.092308 f := 261.3471078 T := 0.003818536004
Finally, the vibrations of the bar are animated over the interval t=0 to 10T .

> animate(psi,x=0..L,t=0..10*T,frames=100,thickness=2);

Execute the recipe on your computer and click on the plot and on the start
arrow to see the vibrations.

4.2.7 A Poisson Recipe

Life is good for only two things, discovering mathematics
and teaching mathematics
Simon Poisson, French mathematician, (1781–1840)

In this day and age, being overweight and having high blood pressure and/or
high cholesterol cause great concern among an aging population. To address
these issues, all types of diets are proposed and cookbooks written. Amongst
the latter are the HeartSmart [Ste94] cook books, sponsored by the Heart and
Stroke Foundation of Canada. Each book presents “over 200 healthful & deli-
cious recipes”, with a section devoted to fish and seafood. So, in the spirit of
keeping us intellectually healthy, here’s a delicious Poisson (equation) recipe.

The simplest non-trivial Poisson equation problems involve point or line
charges in the vicinity of one or more “grounded” (zero potential) conducting
surfaces. Since the charge densities for these sources may be characterized
by Dirac δ-functions, the solutions of these problems are electrostatic Green’s
functions. As a simple 2-dimensional example, let us consider an infinitely long

line charge characterized by the charge density ρ=4 π ε0 δ(x − a) δ(y − b), with
ε0 the permittivity of free space. The line charge is located between two infinite
grounded conducting plates at y =0 and at L > b > 0. To firmly establish the
geometry in our minds, a finite portion of the two infinite conducting planes is
plotted in the x-y plane as green lines in gr1, with L=1.

> restart: with(plots):

> gr1:=plot([[[-1/2,0],[3/2,0]],[[-1/2,1],[3/2,1]]],color=

green,thickness=2,tickmarks=[4,3],labels=["x","y"]):
Taking a = 0.5 and b = 0.8, a point is plotted in gr2, representing an end-on
view of the line charge. The geometry for the Green’s function problem is then
displayed by superimposing the two graphs in Figure 4.9.

> gr2:=pointplot([[0.5,0.8]],symbol=circle,symbolsize=12):

> display({gr1,gr2},scaling=constrained);

Figure 4.9: End-on view of line charge between two grounded conducting plates.

The relevant Poisson equation for the Green’s function potential G is

∇2G =
∂2G

∂x2 +
∂2G

∂y2 = − ρ

ε0
= − 4 π δ(x − a) δ(y − b). (4.7)

For the region outside the line charge, Equation (4.7) becomes homogeneous
and a solution is easily constructed. In the y direction G is zero at y = 0 and
L, so G must include terms of the structure sin(nπ y/L), with n = 1, 2, 3, ....
As x → ±∞, we must have G → 0. For x < a, the Green’s function GL to the
“left” of the line charge will be built up of terms of the structure en π (x−a)/L.
We have used the fact that our final result should ultimately depend only on
the difference x−a. The factor nπ/L is included so that the homogeneous form
of Poisson’s equation will be satisfied. The nth term in the infinite series then
takes the following form for GL, the coefficients An still to be determined.

> GL:=A[n]*sin(n*Pi*y/L)*exp(n*Pi*(x-a)/L);

GL := An sin(
nπ y

L
) e(

n π (x−a)
L )

For x > a, the Green’s function GR to the “right”of the line charge is built up
of terms of the structure e−n π (x−a)/L. The nth term in the infinite series then
takes the following form for GR, the coefficients Bn still to be determined.

> GR:=B[n]*sin(n*Pi*y/L)*exp(-n*Pi*(x-a)/L);



4.2. BEYOND THE STRING 157

GR := Bn sin(
nπ y

L
) e(−

n π (x−a)
L )

As a check on, e.g., GL, we confirm that ∂2(GL)/∂x2 + ∂2(GL)/∂y2 =0.

> check:=diff(GL,x,x)+diff(GL,y,y);

check := 0
The Green’s function must be continuous, i.e., GL=GR, at x=a for arbitrary
y between 0 and L. The continuity condition is implemented in eq1 .

> eq1:=expand(eval(GL=GR,x=a)/sin(n*Pi*y/L));

eq1 := An = Bn

To determine the discontinuity in ∂G/∂x at x=a, we appeal to the divergence
(Gauss’s) theorem, viz.,

∮
V

(∇ · �A) dv =
∮

S
(n̂ · �A) ds for a vector field �A. Here

n̂ is the outward unit normal to the closed surface S, enclosing a volume V .
Let’s take V to be a thin (thickness 2ε, where ultimately ε → 0) slice between
the plates of unit length in the z direction, with faces at x − ε and x + ε, and
�A=∇G. Then, making use of Poisson’s equation, the divergence theorem yields∫ ∫

∇2G dx dy=
∫

(n̂ · ∇G) d�=
∫

∂G

∂n
d�=−4π

∫ ∫
δ(x−a) δ(y−b) dx dy=−4 π.

Here ∂G/∂n is the normal derivative of G and the line integral (
∫

d�) is carried
out along the perimeter of the slice in the x-y plane. As ε → 0, the “ends” of
the slice at y =0 and L will make no contribution to the line integral, so that
limε→0[

∫
along x=a+ε

(∂G/∂x) dy −∫along x=a−ε
(∂G/∂x) dy]=−4π, which implies

(in the limit ε → 0) that (∂G/∂x)a+ε − (∂G/∂x)a−ε =−4πδ(y − b). Multiplying
this result by sin(nπy/L), integrating from y=0 to L, and using orthogonality,
yields the desired second relation for the coefficients. This calculation is imple-
mented in eq2 , the assumptions that b > 0, L > b, and n is an integer being
included to accomplish the integration over the δ-function.

> eq2:=int(eval(diff(GL-GR,x),x=a)*sin(n*Pi*y/L),y=0..L)

=4*Pi*int(Dirac(y-b)*sin(n*Pi*y/L),y=0..L)
assuming b>0,L>b,n::integer;

eq2 :=
1
2

nπ An +
1
2

nπ Bn = 4π sin(
nπ b

L
)

eq1 and eq2 are solved for An and Bn, and the solution assigned.

> sol:=solve({eq1,eq2},{A[n],B[n]}): assign(sol):

Keeping 30 terms in the sum, the complete Green’s function to the left and
right of x=a are given in GL2 and GR2 , respectively.

> GL2:=Sum(GL,n=1..30); GR2:=Sum(GR,n=1..30);

GL2 :=
30∑

n=1

⎛
⎜⎝4 sin(

nπ b

L
) sin(

nπ y

L
) e(

n π (x−a)
L )

n

⎞
⎟⎠

GR2 :=
30∑

n=1

⎛
⎜⎝4 sin(

nπ b

L
) sin(

nπ y

L
) e(−

n π (x−a)
L )

n

⎞
⎟⎠

The parameter values a=0.5, b=0.8, and L=1 are entered, and the piecewise
Green’s function, G=GL2 for x < a and GR2 for x > a, formed.

> a:=0.5: b:=0.8: L:=1:

> G:=piecewise(x<a,value(GL2),x>a,value(GR2)):

Using contourplot, equipotentials are drawn in cp for G=1/8, 1/4, 1/2, 1, 2, 3
and then superimposed on gr1 and gr2, the result being shown in Fig. 4.10.
The curve closest to the line charge is for G=3, the furthest for G=1/8.

> cp:=contourplot(G,x=-2..2,y=0..L,contours=[1/8,1/4,1/2,

1,2,3],grid=[70,70],color=blue,thickness=2):
> display({gr1,gr2,cp},scaling=constrained);

Figure 4.10: Equipotentials around line charge between grounded plates.

4.3 Beyond Cartesian Coordinates

Some representative non-Cartesian examples are now presented. Many more
are included in the Supplementary Recipes at the end of the chapter.

4.3.1 Is It Separable?

Alliance. In international politics, the union of two thieves who have
their hands so deeply inserted in each other’s pockets that they can-
not separately plunder a third.
Ambrose Bierce, American author, The Devil’s Dictionary (1881–1906)

The method of separation of variables can be applied to other orthogonal curvi-
linear coordinate systems besides the Cartesian system. I have found that, at
first, some students are surprised that the variable separation method works



4.3. BEYOND CARTESIAN COORDINATES 159

at all. After a while, they assume that it always works. It turns out that the
scalar Helmholtz equation, ∇2S+k2 S =0, which is the spatial part of either the
wave or diffusion equations with k a constant, is separable [MF53] in 11, and
only 11, 3-dimensional orthogonal curvilinear coordinate systems. Fortunately,
these include spherical polar and cylindrical coordinates, which are the two
most commonly used non-Cartesian systems. An example of a 3-dimensional
coordinate system for which the Helmholtz equation is not separable are the
bispherical coordinates u, v, w, which are related to x, y, z by

x =
a sin u cos v

cosh w − cos u
, y =

a sin u sin v

cosh w − cos u
, z =

a sinhw

cosh w − cos u
. (4.8)

Here a is a scale factor and 0 ≤ u < π, 0 ≤ v ≤ 2 π, −∞ < w < ∞.
As the following recipe illustrates, Laplace’s equation is not separable in

bispherical coordinates either, but can be separated into three ODEs by a mod-
ified separation assumption. This is useful, e.g., in determining the potential
outside two spheres of equal diameters, held at different potentials, and with
their centers separated by a distance greater than the sphere diameter.2

Although, the bispherical system is known (with a = 1) to Maple, it is
instructive to tackle the following problem from first principles.

(a) Plot the contours in the x-z plane corresponding to holding u and w fixed.
What surfaces are generated if v is constant?

(b) Calculate the scale factors and the Laplacian operator.
(c) Show that Laplace’s equation is not completely separable if one makes

the “standard” ansatz, S(u, v, w)=U(u) V (v) W (w).
(d) Show that Laplace’s equation is completely separable if one assumes that

S(u, v, w)=
√

(cosh w − cos u) U(u) V (v) W (w). Assuming that cosh w >
cos u, identify any special functions which occur in the separated ODEs.

It is assumed that u ≥ 0, u < π, v ≥ 0, v ≤ 2π, and cosh w > cos u. The
coordinate relations are then entered.

> restart: with(plots): assume(u>=0,u<Pi,v>=0,v<=2*Pi,

cosh(w)>cos(u)):

> x:=a*sin(u)*cos(v)/(cosh(w)-cos(u));

> y:=a*sin(u)*sin(v)/(cosh(w)-cos(u));

> z:=a*sinh(w)/(cosh(w)-cos(u));

To plot the surfaces corresponding to holding w fixed, let’s form X2 + Y 2 +
(Z − a coth w)2 =x2 + y2 + (z − a coth w)2 and simplify the right-hand side.

> eq1:=Xˆ2+Yˆ2+(Z-a*coth(w))ˆ2

=simplify(xˆ2+yˆ2+(z-a*coth(w))ˆ2);

eq1 := X2 + Y 2 + (Z − a coth(w))2 =
a2

sinh(w)2
The result is the equation of a sphere of radius a/ sinhw centered at X = 0,

2An excellent source of electrostatic problems in bispherical and other coordinate systems
is Problems in Mathematical Physics by Lebedev, Skal’skaya, and Uflyand (Pergamon, 1966).

Y = 0, and Z = a coth w. Different choices of w will generate different size
spheres with centers located at different Z values.

Similarly, (
√

X2 + Y 2−a cot u)2+Z2 =(
√

x2 + y2−a cot u)2+z2 is entered
in eq2 and simplified.

> eq2:=(sqrt(Xˆ2+Yˆ2)-a*cot(u))ˆ2+Zˆ2

=simplify((sqrt(xˆ2+yˆ2)-a*cot(u))ˆ2+zˆ2,symbolic);

eq2 := (
√

X2 + Y 2 − a cot(u))2 + Z2 =
a2

sin(u)2
To see what type of surface is generated by holding u fixed, let’s set a=1 and
Y =0 so as to generate plots in the X-Z plane. The unapply command is used
to free up w and u in eq1 and eq2 for plotting purposes.

> a:=1: Y:=0: A:=unapply(eq1,w): B:=unapply(eq2,u):

Using A and B and the implicitplot command, representative contours are
drawn for fixed w (solid circles) in gr1 and fixed u (dashed circles) in gr2

> gr1:=implicitplot({seq(A(n/2),n=-4..-1),seq(A(n/2),n=1..4)},
X=-5*a..5*a,Z=-5*a..5*a,grid=[70,70],color=red,thickness=2):

> gr2:=implicitplot({seq(B(0.2*Pi*n),n=1..4)},X=-5*a..5*a,
Z=-5*a..5*a,grid=[70,70],color=blue,thickness=2,linestyle=2):

and superimposed to produce Fig. 4.11.

> display({gr1,gr2},scaling=constrained);

Figure 4.11: Contours for fixed w (solid circles) and fixed u (dashed).

The corresponding 3-dimensional surfaces result on rotating the figure about
the vertical (Z) axis, the circles becoming spheres, hence the name “bispherical”
for the coordinate system. From the defining relations between the bispherical

and Cartesian systems, one has y/x=sin v/ cos v=tan v, so holding v fixed will
produce half-planes passing through the z-axis.

A functional operator L is created for generating the Laplacian of S(u, v, w)
on specifying the coordinates u, v, w and the scale factors hu, hv, hw.

> L:=(u,v,w,hu,hv,hw)->(diff(hv*hw*diff(S(u,v,w),u)/hu,u)

+ diff(hu*hw*diff(S(u,v,w),v)/hv,v)
+ diff(hu*hv*diff(S(u,v,w),w)/hw,w))/(hu*hv*hw):

An operator H is formed for producing and simplifying the scale factors.
> H:=u->simplify(sqrt(diff(x,u)ˆ2+diff(y,u)ˆ2+diff(z,u)ˆ2),

symbolic):
Using H the three scale factors are explicitly calculated.

> h[u]:=H(u); h[v]:=H(v); h[w]:=H(w);

hu :=
1

cosh(w) − cos(u)
hv :=

sin(u)
cosh(w) − cos(u)

hw :=
1

cosh(w) − cos(u)
Then employing L, Laplace’s equation, ∇2S(u, v, w)=0, is generated.

> Lap:=L(u,v,w,h[u],h[v],h[w])=0;

Lap :=

(
cos(u)(

∂

∂u
S (u, v, w))

cosh(w) − cos(u)
+

sin(u)2(
∂

∂u
S (u, v, w))

(cosh(w) − cos(u))2
+

sin(u)(
∂2

∂u2 S (u, v, w))

cosh(w) − cos(u)

+

∂2

∂v2 S (u, v, w)

(cosh(w) − cos(u)) sin(u)
+

sin(u) (
∂

∂w
S (u, v, w)) sinh(w)

(cosh(w) − cos(u))2

+
sin(u) (

∂2

∂w2 S (u, v, w))

cosh(w) − cos(u)

)
(cosh(w) − cos(u))3/sin(u) = 0

An unsuccessful attempt is made to separate Laplace’s equation by assuming
that S(u, v, w)=U(u) V (v) W (w).

> pdsolve(Lap,HINT=U(u)*V(v)*W(w));

Warning : Incomplete separation.

(S (u, v, w) = V (v) F1 (u, w)) &where[{ d2

dv2 V (v) = c2 V (v), · · · · · ·

The separation is incomplete, an ODE resulting for V (v), but the u and w de-
pendence remaining coupled in a PDE, which is not displayed here in the text.
Supplying the modified separation assumption as a hint, Laplace’s equation is
now completely separated.

> pdsolve(Lap,HINT=sqrt((cosh(w)-cos(u)))*U(u)*V(v)*W(w),

INTEGRATE);

(S (u, v, w) =
√

cosh(w) − cos(u)U (u)V (v)W (w)) &where

[{{W (w) = C5 e(
√

c3 w) + C6 e(−√
c3 w)},

{V (v) = C3 e(
√

c2 v) + C4 e(−√
c2 v)},

{U (u) = C1 (
1
2

cos(2 u) − 1
2
)(1/2 I

√
c2) sin(2u)hypergeom([

1
2

I
√

c2

+
1
2

√
c3 +

3
4
,

1
2

I
√

c2 − 1
2

√
c3 +

3
4
], [

3
2
],

1
2

cos(2 u) +
1
2
)
/

√
1 − cos(2 u) + · · ·}}]

W (w) and V (v) are both expressed in terms of exponentials, but U is given in
terms of hypergeometric functions. The hypergeometric function F (a, b; c; z) is
given by the following infinite series [AS72], where Γ is the Gamma function,

F (a, b; c; z) =
Γ(c)

Γ(a) Γ(b)

∞∑
n=0

Γ(a + n) Γ(b + n)
Γ(c + n)

zn

n!
.

4.3.2 A Shell Problem, Not a Shell Game

Insurance. An ingenious modern game of chance in which the player
is permitted to enjoy the comfortable conviction that he is beating the
man who keeps the table.
Ambrose Bierce, American author, The Devil’s Dictionary (1881–1906)

In creating physics exams at the freshman level, I often play a bit of a shell
game, presenting problems similar to those that the students have solved for
homework, but in new guises and combinations. The hope is that they really
understand the underlying principles and methods and haven’t merely memo-
rized the solutions to the homework problems. At the senior level, I rely less
on “disguise ” and more on having students explore challenging problems, even
“standard” ones, in some depth. A computer algebra approach is encouraged
as an auxiliary tool. The following recipe, submitted by Ms. I. M. Curious, is
based on a standard problem appearing on an exam given to my senior electro-
magnetic theory class.

A very long circular cylindrical shell of dielectric constant ε and inner and
outer radii a and b, respectively, is placed in a previously uniform electric field
�E0 with the cylinder axis perpendicular to the field. The medium inside (r < a)
and outside (r > b) the cylindrical shell has a dielectric constant of unity.

(a) Determine the potential and electric field in the three regions.

(b) Taking ε=3, a=1, b=2, and E0 =1, plot the equipotentials and electric
field vectors in all three regions in a single figure. Discuss the results and
explore the effect of changing the parameter values.

In addition to the plots library package, I. M. loads the plottools and VectorCal-
culus packages. Plottools contains the circle command which she will use for
drawing the inner and outer radii of the cylindrical shell. The VectorCalculus
package is needed for the Laplacian and Gradient commands.

> restart: with(plots): with(plottools): with(VectorCalculus):

Neglecting end effects, the cylindrical shell is taken to be infinitely long in
the z direction, thus reducing the problem to 2 dimensions in the x-y plane.
Noting the circular symmetry, I. M. introduces the polar coordinates (r, θ) with
x=r cos θ and y=r sin θ, r being measured from the cylinder axis and θ from
the x-axis. Laplace’s equation is entered in polar coordinates and expanded.

> pde:=expand(Laplacian(phi(r,theta),’polar’[r,theta]))=0;

pde :=

∂

∂r
φ(r, θ)

r
+ (

∂2

∂r2 φ(r, θ)) +

∂2

∂θ2 φ(r, θ)

r2 = 0

Using the separation of variables method, a general solution is sought of the
form φ(r, θ)=R(r) Θ(θ). For convenience, I. M. replaces the separation constant√

c1 that appears on the rhs of sol with the symbol k.
> sol:=pdsolve(pde,HINT=R(r)*Theta(theta),INTEGRATE,build);

> sol2:=subs(sqrt(_c[1])=k,rhs(sol));

sol2 := C3 sin(k θ) C1 rk +
C3 sin(k θ) C2

rk
+ C4 cos(k θ) C1 rk

+
C4 cos(k θ) C2

rk

Taking the electric field to be in the x direction, I. M. notes that the solution
must have reflection symmetry (is unchanged if θ → −θ) around the x axis. So
she removes the sine terms, which are odd functions of θ, from sol2 . As r → ∞,
the electric field must remain uniform and is given by �E0 =E0 êx =−(∂φ/∂x) êx,
so the asymptotic potential is φ=−E0 x=−E0 r cos θ, the arbitrary constant in
the potential being set equal to zero. This immediately implies that k=1, which
must hold in every region to satisfy the boundary conditions. I. M. substitutes
k = 1 and, without loss of generality, also sets the redundant coefficient C4
equal to 1 as well.

> sol3:=subs({_C4=1,k=1},remove(has,sol2,sin));

sol3 := cos(θ) C1 r +
cos(θ) C2

r
I. M. labels φ in the regions r < a, a < r < b, and r > b as φ1, φ2, and φ3. An
operator P is formed for relabeling the coefficients C1 and C2 for each φ.

> P:=(u,v)->subs({_C1=u,_C2=v},sol3):
For r < a, the 1/r term must be removed from φ1 for it to remain finite at
r =0. So I. M. forms φ1 by setting v =0 in P and u=A1. For φ2, she chooses
u = A2, v = B2. For φ3, she takes u = −E0, in order to match the asymptotic
boundary condition as r → ∞, and v=B3.

> phi1:=P(A[1],0); phi2:=P(A[2],B[2]); phi3:=P(-E0,B[3]);

φ1 := cos(θ) A1 r

φ2 := cos(θ) A2 r +
cos(θ) B2

r

φ3 := −cos(θ)E0 r +
cos(θ) B3

r
With A1, A2, B2, and A3 unknown, 4 boundary conditions are required. The
potentials φ1 = φ2 at r = a and φ2 = φ3 at b for arbitrary θ. The following
operator F is created to match the potentials u and v at a radius r=R.

> F:=(u,v,R)-> expand(eval((u=v)/cos(theta),r=R)):

Using F, the above boundary conditions are applied in eq1 and eq2 .
> eq1:=F(phi1,phi2,a); eq2:=F(phi2,phi3,b);

eq1 := A1 a = A2 a +
B2

a

eq2 := A2 b +
B2

b
= −E0 b +

B3

b
The radial component of the displacement vector �D = ε �E is continuous at the
boundaries, so ∂φ1/∂r = ε (∂φ2/∂r) at r =a and ε (∂φ2/∂r)=∂φ3/∂r at r = b.
An operator G is created for equating the radial derivative of u and v at a radius
R. Using G, the two boundary conditions are applied in eq3 and eq4 .

> G:=(u,v,R)->expand(eval(diff(u=v,r),r=R)/cos(theta)):

> eq3:=G(phi1,epsilon*phi2,a); eq4:=G(epsilon*phi2,phi3,b);

eq3 := A1 = εA2 − εB2

a2

eq4 := εA2 − εB2

b2 = −E0 − B3

b2

The four equations are solved for the four coefficients and sol4 assigned.
> sol4:=solve({eq1,eq2,eq3,eq4},{A[1],A[2],B[2],B[3]});

assign(sol4):
The potentials φ1, φ2, and φ3 are now determined, the coefficients being auto-
matically substituted.

> phi1:=phi1; phi2:=simplify(phi2); phi3:=phi3;

φ1 := − 4 cos(θ) εE0 b2 r

2 ε a2 − a2 + 2 ε b2 + b2 − ε2 a2 + b2 ε2

φ2 := − 2 cos(θ)E0 b2 (r2 ε + r2 + ε a2 − a2)
(2 ε a2 − a2 + 2 ε b2 + b2 − ε2 a2 + b2 ε2) r

φ3 := −cos(θ)E0 r +
cos(θ)E0 b2 (−ε2 a2 + b2 ε2 + a2 − b2)

(2 ε a2 − a2 + 2 ε b2 + b2 − ε2 a2 + b2 ε2) r
An operator EF is formed for calculating the electric field in polar coordinates,
given some potential f .

> EF:=f->-Gradient(f,’polar’[r,theta]):

The electric field is then explicitly calculated in each region, but only the field
EF1 in region 1 is displayed here.

> EF1:=EF(phi1); EF2:=EF(phi2); EF3:=EF(phi3);

EF1 :=
4 cos(θ) εE0 b2

2 ε a2 − a2 + 2 ε b2 + b2 − ε2 a2 + ε2 b2 er

− 4 sin(θ) εE0 b2

2 ε a2 − a2 + 2 ε b2 + b2 − ε2 a2 + ε2 b2 eθ

The parameter values ε=3, a=1, b=2, and E0 =1 are now entered.
> epsilon:=3: a:=1: b:=2: E0:=1:

For plotting purposes, I. M. changes to Cartesian coordinates by entering r =√
x2 + y2, cos θ=x/r and sin θ=y/r.

> r:=sqrt(xˆ2+yˆ2): cos(theta):=x/r: sin(theta):=y/r:

The potential V for all three regions is formed with the piecewise command
and the radical expressions simplified with the radsimp command. The electric
field Ef is then calculated from V using the Gradient command and again the
radicals are simplified.

> V:=radsimp(piecewise(r<a,phi1,r<b,phi2,phi3));

Ef:=-radsimp(Gradient(V,[x,y]));

V :=

⎧⎪⎪⎪⎪⎪⎨
⎪⎪⎪⎪⎪⎩

−4 x

5

√
x2 + y2 < 1

−4 (2 x2 + 2 y2 + 1)x

15 (x2 + y2)

√
x2 + y2 < 2

− (5 x2 + 5 y2 − 8) x

5 (x2 + y2)
otherwise

The equipotentials of V are produced in gr1 over the range x=−4..4, y=−4..4
with the contourplot command, 25 contours being requested. I. M. colors the
plot by including filled=true as an option.

> gr1:=contourplot(V,x=-4..4,y=-4..4,contours=25,filled=true):

An operator C is formed to plot a thick red circle of radius r, centered at the
origin. Then C is used in gr2 and gr3 to produce circles of radius a and b,
representing the inner and outer radii of the cylindrical shell.

> C:=r->circle([0,0],r,color=red,thickness=3):

gr2:=C(a): gr3:=C(b):
The fieldplot command is used in gr4 to plot the electric field vectors as
medium sized blue arrows. The density of the arrows is controlled.

> gr4:=fieldplot([Ef[1],Ef[2]],x=-4..4,y=-4..4,color=blue,

arrows=MEDIUM,grid=[20,20]):
The four graphs are superimposed with the display command, the scaling
being constrained. The resulting picture is shown in Figure 4.12.

> display({gr1,gr2,gr3,gr4},scaling=constrained);

–4

4

y

–4 4x

Figure 4.12: The electric field arrows and equipotentials for a cylindrical dielec-
tric shell inserted in a previously uniform (horizontal) electric field.

Referring to the figure, I. M. notes that the electric field for r < a is completely
horizontal and therefore parallel to the asymptotic field. The arrows are slightly
shorter however. Quantitatively, the ratio of the electric field for r < a to the
asymptotic field is A1/E0 = 4/5. Unlike the situation for a conductor, the
surfaces of the dielectric shell are not equipotentials so the electric field vectors
are not perpendicular to the inner and outer surfaces of the shell.

I. M. leaves it to you, the reader, to explore her recipe. For example, you
might take ε to be much larger, or alter b with a fixed. She reminds you that
in interpreting any result to remember that the arrows are not field lines.

4.3.3 The Little Drummer Boy

Shall I play for you! pa rum pum pum on my drum.
From the Christmas carol, Little Drummer Boy

Little Daniel loves to bang on large pots and pans with a wooden spoon, creat-
ing his own version of music, which to untrained adult ears such as mine sounds
a lot like noise. Here’s a quieter version of “drum playing”.

A large circular drumhead of radius r = a = 1 m has its perimeter fixed.
If the drumhead has an initial shape U(r, θ, 0) = r (1 − r2/a2) sin(2 θ)/20 and
is released from rest, determine the shape of the drumhead at arbitrary time

t > 0. Take the wave speed to be c=1 m/s. Then, animate the motion of the
drumhead in time steps of 0.1 s over the interval t=0 to 2 s.

After loading the plots and VectorCalculus packages, the wave equation pde
is entered in polar coordinates (r, θ), making use of the Laplacian command.

> restart: with(plots): with(VectorCalculus):

> pde:=expand(Laplacian(U(r,theta,t),’polar’[r,theta]))

=(1/cˆ2)*diff(U(r,theta,t),t,t);

pde :=

∂

∂r
U (r, θ, t)

r
+ (

∂2

∂r2 U (r, θ, t)) +

∂2

∂θ2 U (r, θ, t)

r2 =

∂2

∂t2
U (r, θ, t)

c2

Then pde is analytically solved, assuming that U(r, θ, t)=R(r) Θ(θ) T (t).
> sol:=pdsolve(pde,HINT=R(r)*Theta(theta)*T(t),INTEGRATE,

build);

sol := U (r, θ, t) = e(
√

c3 t) e(
√

c2 θ) C5 C3 C1 BesselJ(
√− c2,

√− c3 r

c
)

+ e(
√

c3 t) e(
√

c2 θ) C5 C3 C2 BesselY(
√− c2,

√− c3 r

c
) + · · ·

The answer involves exponentials and Bessel functions of the first and second
kinds. The separation constants c2 and c3 on the rhs of sol are replaced with
−p2 and −c2 k2, and the result simplified with the symbolic option.

> U1:=simplify(subs({_c[2]=-pˆ2,_c[3]=-cˆ2*kˆ2},rhs(sol)),
symbolic);

U1 := C5 C3 C1 BesselJ(p, k r) e((c k t+p θ) I)

+ C5 C3 C2 BesselY(p, k r) e((c k t+p θ) I) + · · ·
The Bessel functions Yp(k r) of the second kind diverge at r=0, so are removed
from U1 . Then U2 is converted to trig form and expanded in U3 .

> U2:=remove(has,U1,BesselY);

> U3:=expand(convert(U2,trig));

U3 := C5 C3 C1 BesselJ(p, k r) cos(c k t) cos(p θ)
− C5 C3 C1 BesselJ(p, k r) sin(c k t) sin(p θ)
+ C5 C4 C1 BesselJ(p, k r) sin(c k t) cos(p θ) I

+ C5 C3 C1 BesselJ(p, k r) cos(c k t) sin(p θ) I + · · ·
The initial transverse velocity of the drumhead is zero, so the sin(c k t) terms
must be removed from U3 . Since the initial shape involves a sine function only,
there can be no cosine terms present in the solution. The cos(p θ) terms are
therefore also removed in U4 and the result then factored.

> U4:=factor(remove(has,U3,{sin(c*k*t),cos(p*theta)}));
U4 := C1 ( C3 − C4 ) ( C5 + C6 ) sin(p θ) cos(c k t) BesselJ(p, k r)I

The solution must vanish at the perimeter, so the Bessel functions Jp(k r) must
be zero at r = a. Thus, k a must equal the mth zero of Jp, with m = 1, 2, ....
The allowed k values are now entered.

> k:=BesselJZeros(p,m)/a:

The parameter values a = 1 and c = 1 are given. To match the initial angular
dependence, we must have p=2, i.e., only the Bessel function J2 will be present
in the solution. The radial portion of the initial shape is entered in f . The
form of the Bessel functions is given in g.

> a:=1: c:=1: p:=2: f:=r*(1-rˆ2/aˆ2)/20; g:=BesselJ(p,k*r);

f :=
r (−r2 + 1)

20
g := BesselJ(2, BesselJZeros(2, m) r)

The messy constants are removed from U4 with the following select command.
> U[2,m]:=select(has,U4,{cos,sin,BesselJ});

U2, m := cos(BesselJZeros(2, m) t) sin(2 θ) BesselJ(2, BesselJZeros(2, m) r)
Making use of orthogonality and noting that the weight function for the Bessel
functions is r, the mth coefficient is given by A2,m =

∫ a

0 f r g dr/
∫ a

0 r g2 dr.
> A[2,m]:=int(f*r*g,r=0..a)/int(r*gˆ2,r=0..a):

The first 4 terms of the series solution U are now displayed in decimal form.
> U:=evalf(sum(A[2,m]*U[2,m],m=1..4));

U := 0.04223579864 cos(5.135622302 t) sin(2. θ) BesselJ(2., 5.135622302 r)
+ 0.002159293112 cos(8.417244140 t) sin(2. θ) BesselJ(2., 8.417244140 r)
+ 0.005912588156 cos(11.61984117 t) sin(2. θ)BesselJ(2., 11.61984117 r)
+ 0.001008506062 cos(14.79595178 t) sin(2. θ) BesselJ(2., 14.79595178 r)

For animation purposes, we convert from polar coordinates to Cartesian coordi-
nates, setting r=

√
x2 + y2 and using the trig identity sin(2 θ)=2 sin θ cos θ=

2 x y/r2. Note that a floating point evaluation is used in entering the latter so
that the substitution will actually occur. This is necessary because a floating
point evaluation was used in expressing U .

> r:=sqrt(xˆ2+yˆ2): sin(evalf(2*theta)):=2*x*y/rˆ2:

The solution is then expressed as the piecewise function UU =U for r < a and
0 for r > a.

> UU:=evalf(piecewise(r<a,U,r>a,0)):

To animate UU , a functional operator gr is created to make a 3-dimensional
plot of the drumhead shape on the ith time step, the stepsize being 0.1 s.

> gr:=i->plot3d(eval(UU,t=0.1*i),x=-a..a,y=-a..a,

style=patchcontour,shading=zhue):
Then using the sequence command, the profiles on 20 consecutive time steps are
displayed. The insequence=true option is included to produce the animation.

> display([seq(gr(i),i=1..20)],insequence=true,axes=framed);

If you wish to see the drumhead animation, execute the recipe on your computer,
then click on the computer plot and on the start arrow in the tool bar.

4.3.4 The Cannon Ball

The sound of a kiss is not so loud as that of a cannon,
but its echo lasts a great deal longer.
Oliver Wendell Holmes Sr., American writer, physician, (1809–94)

In the re-enactment of a Civil war battle, a cannon is fired and a hot spheri-
cal iron cannon ball plunges into an icy lake whose temperature is very close
to freezing (0◦C). If the cannon ball has a radius R = 20 cm and is initially
100◦C throughout on entering the lake, determine the temperature distribu-
tion inside the cooling cannon ball as a function of time. Plot the temperature
distribution in 1 minute intervals up to 15 minutes. What is the temperature
at the center of the cannon ball 15 minutes after plunging into the lake? For
iron, K/(ρC) = 0.185 in cgs units, where K is the thermal conductivity, ρ the
density, and C the specific heat.

After loading the plots and VectorCalculus packages,

> restart: with(plots): with(VectorCalculus):

the heat diffusion equation ∇2T =(1/a2) (∂T/∂t), with T the temperature and
a2 ≡ K/(ρ C), is entered in spherical polar coordinates (r, θ, φ). r is the radial
distance from the center of the cannon ball, θ is the angle that the radial vector
makes with the z-axis, and φ is the angle that the projection of the radial vector
into the x-y plane makes with the x-axis.

> pde:=expand(Laplacian(T(r,theta,phi,t),’spherical’))

[r,theta,phi]=diff(T(r,theta,phi,t),t)/aˆ2;

pde :=
2 (

∂

∂r
T (r, θ, φ, t))

r
+(

∂2

∂r2 T (r, θ, φ, t)) +
cos(θ) (

∂

∂θ
T (r, θ, φ, t))

r2 sin(θ)

+

∂2

∂θ2 T (r, θ, φ, t)

r2 +

∂2

∂φ2 T (r, θ, φ, t)

r2 sin(θ)2
=

∂

∂t
T (r, θ, φ, t)

a2

Since the initial temperature of the cannon ball is uniform throughout, the so-
lution will have no angular dependence. Assuming that T (r, θ, φ, t)=R(r) F (t),
the heat flow equation pde is analytically solved using the pdsolve command.

> sol:=pdsolve(pde,HINT=R(r)*F(t),INTEGRATE,build);

sol := T (r, θ, φ, t) =
C3 e(a2 c1 t) C1 sinh(

√
c1 r)

r

+
C3 e(a2 c1 t) C2 cosh(

√
c1 r)

r
The hyperbolic sine (sinh) term remains finite at the origin (r=0), but the cosh
term diverges to ∞ and must be removed from the rhs of sol .

> T1:=remove(has,rhs(sol),cosh);

T1 :=
C3 e(a2 c1 t) C1 sinh(

√
c1 r)

r
The separation constant c1 is replaced with −k2 in T1 and the result simplified
with the symbolic option.

> T2:=simplify(subs(_c[1]=-kˆ2,T1),symbolic);

T2 :=
C3 e(−a2 k2 t) C1 sin(k r) I

r
It should be noted that the term sin(k r)/r is ([AS72]) just the zeroth order
spherical Bessel function of the first kind.3 Making use of the select command,
the coefficient combination C3 C1 I in T2 is replaced with the symbol A.

> T3:=A*select(has,T2,{exp,sin,r});

T3 :=
A e(−a2 k2 t) sin(k r)

r
Taking the radius of the cannon ball to be R, the surface of the ball is held at
0 degrees, so sin(k R)=0, and therefore k =nπ/R, with n=1, 2, .... Entering
this result, the nth normal mode of the temperature is displayed in T4 .

> k:=n*Pi/R: T4:=T3;

T4 :=
A e(−a2 n2 π2 t

R2 ) sin(
nπ r

R
)

r
The initial temperature is 100◦ throughout the cannon ball. Noting that the
weight function ([AS72]) for the spherical Bessel functions is r2, orthogonality
leads to the following expression for the coefficients:

A=
∫ R

0
r2 100 (sin(k r)/r) dr/

∫ R

0
r2 (sin(k r)/r)2 dr.

A is now calculated, assuming that n is an integer.
> A:=int(rˆ2*100*sin(k*r)/r,r=0..R)/int(rˆ2*(sin(k*r)/r)ˆ2,

r=0..R) assuming n::integer;

A :=
200 (−1)(1+n) R

nπ
The formal series representation of the temperature distribution T inside the
cannon ball is now completely determined. Retaining 200 terms in the series,
T has the following structure.

> T:=Sum(T4,n=1..200);

T :=
200∑
n=1

⎛
⎜⎝200 (−1)(1+n) R e(−a2 n2 π2 t

R2 ) sin(
nπ r

R
)

nπ r

⎞
⎟⎠

3The spherical Bessel functions jn of the first kind are related to the “ordinary” Bessel
functions by the relation jn(x) ≡

√
(π/2x) Jn+1/2(x).

Taking R=20 and a=
√

0.185, T is now evaluated, but not displayed.
> T:=eval(value(T),{R=20,a=sqrt(0.185)}):

An arrow operator gr is formed to plot T at 60 s (1 min) intervals.
> gr:=i->plot(eval(T,t=i*60),r=0..20,numpoints=1000):

Using gr and the sequence command, the temperature T is displayed as a
function of radius r in Figure 4.13 at t=0 (top curve), t=1 min (next lowest
curve), etc, to t=15 minutes (bottom curve).

> display(seq(gr(i),i=0..15),labels=["r","T"]);

0

50

100

T

10 r 20

Figure 4.13: Time evolution of the temperature T inside the cannon ball.

The “ringing” in the initial temperature distribution is, of course, the Gibb’s
phenomenon due to the initial step function temperature profile. Taking the
limit of T as r → 0, the temperature at the center of the sphere at 15 minutes,
or 900 seconds, is calculated and found to be about 3.3◦C.

> Tcenter:=eval(limit(T,r=0),t=900);

Tcenter := 3.287377363

4.3.5 Variation on a Split-sphere Potential

Would you convey my compliments to the purist who reads your
proofs and tell him or her ... that when I split an infinitive, God damn
it, I split it so it will stay split.
Raymond Chandler, American writer of detective fiction, (1888–1959)

The following recipe is based on a simple variation of a standard problem in
electrostatics. The objective is to completely determine the potential V inside
and outside a hollow sphere of unit radius with a specified piecewise potential
on the spherical surface. With θ measured from the positive z-axis (point-
ing vertically upwards), the spherical surface between θ = 0 and 45◦ has the



172 CHAPTER 4. LINEAR PDES OF PHYSICS

constant potential V0 , the intermediate section between 45◦ and 135◦ has a
variable potential given by

√
2 cos(θ)V0 , and the lower portion between 135◦

and 180◦ held at −V0 . Keeping the first 6 non-zero terms in V and taking
V0 =1, plot the equipotentials corresponding to V =0.8, 0.6, ..., −0.6, −0.8.

After loading the necessary library packages, Laplace’s equation is entered in
spherical coordinates, the origin taken at the center of the sphere. By symmetry,
V must be independent of the azimuthal4 angle φ, i.e., V = V (r, θ).

> restart: with(plots): with(VectorCalculus):

> pde:=expand(Laplacian(V(r,theta),’spherical’

[r,theta,phi]))=0;

pde :=
2 (

∂

∂r
V(r, θ))

r
+ (

∂2

∂r2 V(r, θ)) +
cos(θ) (

∂

∂θ
V(r, θ))

r2 sin(θ)
+

∂2

∂θ2 V(r, θ)

r2 = 0

Then pde is analytically solved, assuming that V (r, θ) = R(r) Θ(θ), the result
involving Legendre functions.

> V:=rhs(pdsolve(pde,HINT=R(r)*Theta(theta),INTEGRATE,build));

V :=
C3 LegendreP(−1

2
+

1
2

√
1 + 4 c1,cos(θ)) C1 r(1/2

√
1+4 c1)

√
r

+ · · ·

The separation constant c1 is replaced in V with −1/4 + (n + 1/2)2 and the
result simplified with the symbolic option and then expanded.

> V:=expand(simplify(subs(_c[1]=-1/4+(n+1/2)ˆ2,V),symbolic));

V := C3 LegendreP(n, cos(θ)) C1 rn +
C3 LegendreP(n, cos(θ)) C2

r rn

+ C4 LegendreQ(n, cos(θ)) C1 rn +
C4 LegendreQ(n, cos(θ)) C2

r rn

V is expressed in terms of the Legendre functions of the first (Pn(cos θ)) and
second (Qn(cos θ)) kinds. The Qn diverge at the end points of the θ range and
must be rejected. The redundant constant C3 in the Pn terms is set equal to
1 and C1 and C2 are replaced with the symbols A and B.

> V:=subs({LegendreQ(n,cos(theta))=0,_C3=1,_C1=A,_C2=B},V);

V := LegendreP(n, cos(θ)) A rn +
LegendreP(n, cos(θ)) B

r rn

For the inside (r < 1) solution Vin, we set B = 0 in V so that Vin doesn’t
diverge at the origin. For r > 1, we take A=0 so that Vout → 0 as r → ∞.

> Vin:=subs(B=0,V); Vout:=subs(A=0,V);

Vin := LegendreP(n, cos(θ)) A rn

4The angle between the projection of the radius vector into the x-y (horizontal) plane and
the x-axis.

Vout :=
LegendreP(n, cos(θ)) B

r rn

Setting u=cos θ, the angular distributions in each region are entered.
> f1:=-V0: f2:=sqrt(2)*u*V0: f3:=V0:

Making use of orthogonality of the Pn(u), the coefficients in the series repre-
sentation of the solution are evaluated using An =((2n + 1)/2)

∫ 1
−1 f Pn(u) du.

An operator AA is introduced to evaluate the coefficients for a given n value.
> AA:=n->((2*n+1)/2)*(int(f1*LegendreP(n,u),u=-1..-1/sqrt(2))

+int(f2*LegendreP(n,u),u=-1/sqrt(2)..1/sqrt(2))
+int(f3*LegendreP(n,u),u=1/sqrt(2)..1)):

Then, employing AA(n), the inside and outside solutions are determined, the
series being terminated at n=12.

> VIN:=sum(eval(Vin,A=AA(n)),n=0..12);

VIN :=
5
4

cos(θ)V0 r − 7
32

LegendreP(3, cos(θ))V0 r3

− 11
128

LegendreP(5, cos(θ))V0 r5 +
85

2048
LegendreP(7, cos(θ))V0 r7

+
323
8192

LegendreP(9, cos(θ))V0 r9 − 1219
65536

LegendreP(11, cos(θ))V0 r11

> VOUT:=sum(eval(Vout,B=AA(n)),n=0..12);

VOUT :=
5
4

cos(θ)V0
r2 − 7

32
LegendreP(3, cos(θ))V0

r4 + · · ·

The equipotentials will now be plotted in the x-z plane, by setting r=
√

x2 + z2

and cos θ=z/r. We also set V 0=1.
> r:=sqrt(xˆ2+zˆ2): cos(theta):=z/r: V0:=1.0:

A piecewise potential function VPW is formed with V = VIN for r < 1 and
VOUT for r > 1.

> VPW:=piecewise(r<1,VIN,r>1,VOUT):

Loading the plottools package, a blue circle of radius 1 centered on the origin
is produced in c to represent the spherical surface (a circle in 2 dimensions).

> with(plottools): c:=circle([0,0],1,color=blue,thickness=2):

The contourplot command is used in cp to plot the requested equipotentials.
> cp:=contourplot(VPW,x=-3..3,z=-4..4,contours=

[seq(0.8-0.2*i,i=0..8)],grid=[60,60],thickness=2,color=red):
The graphs c and cp are now displayed together with constrained scaling,

> display({c,cp},scaling=constrained);
the resulting picture being shown in Figure 4.14.

Figure 4.14: Equipotentials for the split-sphere in the x-z plane.

The 3-dimensional equipotential surfaces are obtained by mentally rotating the
picture about the z-axis. The recipe is easily adjusted to handle other variations
on the angular distribution and can even be modified to handle cylindrical
geometry.

4.3.6 Another Poisson Recipe

Every man is a potential genius until he does something.
Sir Herbert Beerbohm Tree, English actor-manager, (1853–1917)

In magnetostatics, the vector potential �A(�R ) at a point �R, associated with
a current density �J in free space, satisfies the vector Poisson equation, [Gri99]

∇2 �A = −µ0 �J, (4.9)

where µ0 is the permeability of free space. Equation (4.9) is three scalar Poisson
equations, one for each Cartesian component, e.g., ∇2Ax =−µ0 Jx. In any other
curvilinear coordinate system, the unit vectors are functions of position. Thus,
e.g., in spherical polar coordinates it is not true that ∇2Ar =−µ0 Jr.
Assuming that �J → 0 at infinity, the solution of Eq. (4.9) is given by,

�A(�R ) =
µ0

4 π

∫ �J(�R1)

|�R − �R1|
dv1, (4.10)

again representing three 3-dimensional integrals, e.g.,

Ax(x, y, z)=
µ0

4 π

∫ ∫ ∫
Jx(x1, y1, z1) dx1 dy1 dz1√

(x − x1)2 + (y − y1)2 + (z − z1)2
.

If you want to calculate the integrals in Equation (4.10) in other curvilinear
coordinates, you must first express �J in terms of its Cartesian components.
Once �A is determined, the magnetic field is then given by �B = ∇ × �A. As a
representative example, consider the following magnetostatic problem.

A uniformly charged solid sphere of radius a carries a total charge Q and
is spinning with angular velocity ω about the vertical z-axis. Determine the
magnetic vector potential �A inside and outside the sphere and then calculate
the magnetic field �B in both regions. Plot �B/(µ0Qω) for a=1.

Taking the origin at the sphere’s center, we let �R (r, θ, φ) be the location
of the “observation” point P inside or outside the sphere and �R1 (r1 , θ1, φ1)
be the location of a current density “source” point P1 inside the sphere. The
polar angles θ and θ1 are measured from the z-axis and the azimuthal angles φ
and φ1 from the projection of the radius vector into the (horizontal) x-y plane
with the x-axis. To ensure a later simplification, it is assumed that r > 0.

> restart: with(plots): with(VectorCalculus): assume(r>0):

Since the charge is uniformly distributed in the sphere, the charge density ρ is
equal to the total charge Q divided by the volume 4π a3/3 of the sphere. At a
radial distance r1 and angle θ1 with the z-axis, the source point P1 has a linear
velocity r1 sin(θ1) ω. So the current density magnitude J is equal to ρ times
the linear velocity, the entered form of ρ being automatically substituted. The
corresponding vector �J points in the φ̂ direction for every P1.

> rho:=Q/((4*Pi*aˆ3)/3); J:=rho*r1*sin(theta1)*omega;

ρ :=
3 Q

4 π a3 J :=
3
4

Q r1 sin(θ1) ω

π a3

Without loss of generality, let’s take the observation point P to be in the x-z
plane, so that φ=0. The position vector �R of P is now entered, being expressed
in terms of the Cartesian unit vectors along the x and z axes.

> R:=<r*sin(theta),0,r*cos(theta)>;

R := r sin(θ) ex + r cos(θ) ez

The position vector �R1 of the source point P1 is also entered.
> R1:=<r1*sin(theta1)*cos(phi1),r1*sin(theta1)*sin(phi1),

r1*cos(theta1)>;

R1 := r1 sin(θ1) cos(φ1) ex + r1 sin(θ1) sin(φ1) ey + r1 cos(θ1) ez

To calculate �A, let’s first evaluate 1/(|�R− �R1|)=1/
√

(�R − �R1) · (�R − �R1). This
is done in f using the DotProduct command, the result then being simplified
with the symbolic option.

> f:=simplify(1/sqrt(DotProduct(R-R1,R-R1)),symbolic);

f :=
1√

r2 − 2 r sin(θ) r1 sin(θ1) cos(φ1) + r1 2 − 2 r cos(θ) r1 cos(θ1)
We could attempt to evaluate the integral in �A directly, but it is more instructive
to Taylor expand f about a specified value of r1 , out to some given order, and

see what the various orders contribute to the overall answer. This approach
is equivalent to the multipole expansion discussed in standard electromagnetic
texts such as Griffiths [Gri99]. So, a functional operator T is formed to Taylor
expand f to order n about a specified point r1 =d. The convert( ,polynom)
command is included to remove the order of term which would otherwise appear.

> T:=(n,d)->convert(taylor(f,r1=d,n),polynom):

The current density vector can be resolved into the Cartesian components
Jx = J sin(φ1), Jy = J cos(φ1), and Jz = 0. At the observation point (cho-
sen in the x-z plane), the Jx contribution to �A will add up to zero, but the Jy

contribution will not. But since, our choice of observation point in the x-z plane
was arbitrary and there is complete rotational symmetry about the z-axis, the
resultant component Jy will yield the φ component of �A. A functional operator
A is created to perform the volume integration in (4.10) using Jy and the Taylor
expansion of f and taking spherical polar coordinates. The volume element is
r1 2 sin(θ1) dθ1 dφ1 dr1 . The angular coordinate θ1 ranges from 0 to π, while
φ1 varies from 0 to 2π. The order n, the radial distance d about which Taylor
expansion is taking place, and the lower and upper limits, d1 and d2, of the r1
integration must be specified. Again the result is simplified.

> A:=(n,d,d1,d2)->simplify((mu[0]/(4*Pi))*int(int(int(T(n,d)

*J*cos(phi1)*r1ˆ2*sin(theta1),theta1=0..Pi),r1=d1..d2),
phi1=0..2*Pi),symbolic):

First, let’s take the observation point P to be outside the sphere, i.e., r > a.
Since r1 ≤ a, then r1/r < 1 and we can Taylor expand f about r1 =d=0. The
limits of the r1 integration are d1=0 and d2=a. Making uses of the operator
A with the above arguments, the vector potential is evaluated in Out to order
n=1, 2, and 3 and the result assigned.

> Out:=seq(A||n=A(n,0,0,a),n=1..3); assign(Out):

Out := A1 = 0, A2 =
1
20

µ0 sin(θ) a2 Q ω

π r2 , A3 =
1
20

µ0 sin(θ) a2 Q ω

π r2

For n=1, the so-called monopole contribution A1 to the vector potential is 0,
a well-known general result. For n=2, there is a non-zero dipole contribution
A2 . For n = 3 (and higher), there is no additional contribution to the vector
potential, indicating that outside the sphere the vector potential (and hence,
the magnetic field) is that of a “pure” magnetic dipole. The vector potential
Aout outside the sphere is now expressed in spherical polar coordinates, having
only a φ component.

> Aout:=VectorField(<0,0,A2>,’spherical’[r,theta,phi]);

Aout :=
1
20

µ0 sin(θ) a2 Q ω

π r2 eφ

Now, consider P to be inside the sphere, i.e., r < a. For r1 < r, we can again
Taylor expand f about r1 =0, the integration being from r1 =0 to r. But for
r1 > r, f is Taylor expanded about r1 =∞, the r1 integration being from r to
a. Adding the two contributions, �A inside the sphere is calculated for n=1, 2,

and 3. No further contribution occurs for higher n values.
> In:=seq(AA||n=A(n,0,0,r)+A(n,infinity,r,a),n=1..3);

assign(In):
In := AA1 = 0, AA2 =

1
20

µ0 sin(θ) Q ω r3

π a3 ,

AA3 =
1
20

µ0 sin(θ) Q ω r3

π a3 − 1
8

µ0 Q ω r sin(θ) (−a2 + r2)
π a3

In terms of spherical polar coordinates, the vector potential Ain inside the
sphere takes the following form.

> Ain:=VectorField(<0,0,AA||3>,’spherical’[r,theta,phi]);

Ain := (
1
20

µ0 sin(θ) Q ω r3

π a3 − 1
8

µ0 Q ω r sin(θ) (−a2 + r2)
π a3 ) eφ

Using Curl, the magnetic field is calculated outside and inside the sphere.
> Bout:=Curl(Aout); Bin:=simplify(Curl(Ain));

Bout :=
1
10

µ0 a2 Q ω cos(θ)
r3 π

er +
1
20

sin(θ) µ0 a2 Q ω

r3 π
eθ

Bin := − 1
20

cos(θ) µ0 Q ω (3 r2 − 5 a2)
π a3 er +

1
20

sin(θ) µ0 Q ω (6 r2 − 5 a2)
π a3 eθ

To plot �B/(µ0Qω), the MapToBasis command is used to convert the normalized
magnetic field to Cartesian coordinates.

> F:=u->MapToBasis(u/(mu[0]*Q*omega),’cartesian’[x,y,z]):

Taking a = 1, we will plot the magnetic field in the x-z plane. To accomplish
this, let’s set r =

√
x2 + z2 and form an operator G to evaluate the magnetic

field for y=0.
> a:=1: r:=sqrt(xˆ2+zˆ2): G:=B->simplify(eval(F(B),y=0)):

The magnetic field inside and outside the sphere then takes the following forms,
expressed in Cartesian coordinates.

> Bin2:=G(Bin); Bout2:=G(Bout);

Bin2 :=
3 x z

20 π
ex − 6 x2 + 3 z2 − 5

20 π
ez

Bout2 :=
3 z x

20 (x2 + z2)(5/2) π
ex − −2 z2 + x2

20 (x2 + z2)(5/2) π
ez

The complete magnetic field �B is formed with the piecewise operator PW, taking
the nth component of Bin2 and Bout2 for r < a and r > a, respectively.

> PW:=n->piecewise(r<a,Bin2[n],r>a,Bout2[n]):

The x and z components of �B are obtained by taking n=1 and 3 in PW.
> B[1]:=PW(1): B[3]:=PW(3):

In c, a blue circle of radius a is plotted to represent the spherical surface.
> c:=plot(a,theta=0..2*Pi,coords=polar,color=blue,thickness=2):

The fieldplot command is used in fp to plot the magnetic field as thick red
arrows, the grid density being taken to be 10 × 10.

> fp:=fieldplot([B[1],B[3]],x=-1.1..1.1,z=-1.1..1.1,

arrows=THICK,grid=[10,10],color=red):
The two graphs are superimposed with the display command to produce Fig-
ure 4.15. The picture should be mentally rotated around the z-axis to obtain
the 3-dimensional magnetic field of the rotating uniformly charged sphere.

> display({c,fp});

Figure 4.15: Magnetic field of rotating charged sphere in x-z plane.

Along the spin axis, the magnetic field points vertically upwards, having max-
imum strength at the center of the sphere. The magnetic field rapidly drops
in strength outside the sphere. The arrows form curved paths which leave the
top of the sphere and loop back on the outside to re-enter the sphere at the
bottom. The magnetic field behavior is characteristic of a dipole field.

4.4 Supplementary Recipes
04-S01: General Solutions
Consider a second order linear PDE of the general form (a, b, c are constants
and the subscripts denote derivatives),

a ψxx + b ψxy + c ψyy = 0.

(a) By assuming a solution of the form ψ=f(x+r y) where f is arbitrary, show
that the general solution of the PDE is ψ=f(x+ r1 y)+g(x+ r2 y) where

r1 and r2 are distinct roots of the quadratic equation c r2 + b r + a = 0.
Determine the two roots. If r2 = r1, show that the general solution is
ψ=f(x + r1 y) + y g(x + r1 y), where f and g are arbitrary functions.

(b) Making use of (a), find the general solutions of the following PDEs. In
each case, confirm the solution by solving the PDE directly with pdsolve.

(i) ψxy −ψyy = 0; (ii) ψxx +ψxy −2 ψyy = 0; (iii) ψxx −2 ψxy +ψyy = 0.

04-S02: Balalaika Blues
While trying to create music with a small balalaika, Justine plucks one of the
strings which is initially at rest. If the string is fixed at x=0 and L and has an
initial transverse profile ψ(x, 0)=hx3 sin(3π x/2L) (L−x)2/L5, determine the
subsequent displacement ψ(x, t) of the string using a Fourier series approach.
Taking L=20 cm, h=30 cm, and wave speed c=1 cm/s, animate the motion of
the string. Leave the scaling unconstrained for better viewing of the vibrations.

04-S03: Damped Oscillations
If damping (damping coefficient R) is included, the transverse vibrations of a
light, homogeneous, string under tension are governed by

∂2ψ

∂x2 =
1
c2

∂2ψ

∂t2
+

2R

c2

∂ψ

∂t
.

Consider a horizontal string of length L fixed at x = 0 and L. If it is initially
at rest and is given the initial shape f(x) = hx (x − L), use the separation of
variables method to determine ψ(x, t) for t > 0. Taking L=2 m, h=0.1 m−1,
speed c=5 m/s, and R=1 s−1, animate the motion of the string over the time
interval t=0 to 10 s. You should observe underdamped oscillations. Show that
overdamping occurs for R=10 s−1.

04-S04: Kids Will Be Kids
Consider a very long uniform clothesline under sufficient tension T to keep it
horizontal, with Justine’s recently washed soccer jersey (mass m) attached with
a single clothes peg to the middle of the line at x=0. In an unsuccessful attempt
to dump her sister’s jersey onto the ground, Gabrielle shakes the clothesline in
such a way that a plane wave with amplitude A and phase velocity c=ω/k is
incident from x < 0. Show that reflection and transmission occur at x=0 and
that the energy reflection coefficient R = sin2(θ) and transmission coefficient
Tr=cos2(θ), where θ=arctan(mω2/2kT ). Determine the phase angle changes
for the reflected and transmitted waves in terms of θ.

04-S05: Energy of a Vibrating String
For a finite horizontal string, with linear density ε(x) and under tension T ,
stretched between x=x1 and x2, the kinetic energy for small transverse vibra-
tions is KE=(1/2)

∫ x2

x1
ε(x) (∂ψ/∂t)2 dx. If the ends of the string are either fixed

(ψ=0) or free (ψ ′ =0), the potential energy is PE=(1/2)
∫ x2

x1
T (∂ψ/∂x)2 dx.



180 CHAPTER 4. LINEAR PDES OF PHYSICS

Consider a horizontal string of constant density ε and under tension T fixed
at x=0 and L. If at time t=0, the string is given the initial triangular profile
f(x) = 2h x/L for 0 ≤ x ≤ L, f(x) = 2h (L − x)/L for L/2 ≤ x ≤ L, and is
released from rest, calculate explicit analytic expressions for the kinetic energy,
potential energy, and total energy. What fraction of the total energy is in
the fundamental (lowest) frequency? first harmonic (next highest frequency)?
second harmonic? Plot the kinetic and potential energies divided by the total
energy as a function of time for L=1 m and wave velocity c=

√
T/ε=1 m/s.

04-S06: Vibrations of a Tapered String
A horizontal tapered string fixed at x=0 and L has a linear density ε=a(1+b x).
Determine the normal modes for transverse oscillations of the string. Identify
the functions which occur in the analytic solution. If L=1 m, a=1/1000 kg/m,
b=2 m−1, and the tension T =1 N, determine the three lowest eigenfrequencies.
Animate the fundamental mode over one period, taking 50 frames.

04-S07: Green Function for Forced Vibrations
A stretched (tension T ) string fixed at x = 0 and L is subjected to forced
vibrations by an external force per unit length f(x, t)=F δ(x − ζ) e−i ω t with
the force amplitude F =T and 0 < ζ < L. Solve this Green function problem
and show that the Green function (spatial part) is

G=
sin[k(L − ζ)] sin(k x)

k sin(k L)
, x ≤ ζ, G=

sin[k(L − x)] sin(k ζ)
k sin(k L)

, x ≥ ζ,

where k=ω/v, v being the wave velocity. G has singularities at certain values
of k. Determine these values. Explain, in physical terms, the origin of these
singularities and how in real life, they would be “removed”.

04-S08: Plane-wave Propagation in a 5-Piece String
Consider an infinitely long string which has a linear density ε1 = 1 in region
1 (x < 0), density ε2 = 4 in region 2 (0 < x < L), density ε3 = 1 in region 3
(L < x < 2L), density ε4 = 4 in region 4 (2L < x < 3L), and density ε5 = 1 in
region 5 (x > 3L). For a plane wave of incident amplitude 1, frequency ω, and
wave number K, coming from x = −∞:
(a) Derive the energy transmission coefficient T into region 5 and simplify

the result as much as possible.

(b) Show that T + R = 1, where R is the energy reflection coefficient.

(c) Plot R and T in the same graph as a function of K L up to K L = 6. At
what values of K L is there 100% transmission? Discuss your answer.

04-S09: Transverse Vibrations of a Whirling String
Derive the wave equation for transverse vibrations of a light string of length L
pivoted at one end and whirling in a horizontal plane about that pivot with an
angular velocity ν. Determine the normal modes of oscillation of the whirling
string and the five lowest allowed frequencies. Animate the mode with the
second lowest frequency, taking L=1 m and ν =1 s−1.

04-S10: Newton Would Think That This Recipe Is Cool
Newton’s law of cooling says that the heat flux (amount of heat crossing a unit
area per unit time) across a surface is proportional to the temperature difference
TS − T0 between the surface (temperature TS) and the surrounding medium
(temperature T0). Explicitly this gives the boundary condition n̂ · ∇T |S =
−h(TS − T0), where n̂ is the outward unit normal to the surface and h is the
heat exchange coefficient. Consider an infinitely long uniform bar of rectangular
cross-section with axis along the z direction. The two opposite faces at y=0 and
y = b are held at the temperatures T =0 and T =A, while the other two faces
at x=±a radiate heat according to Newton’s cooling law into the surrounding
medium which is held at T0 =0. Derive the solution which takes the form

T (x, y) =
∞∑

n=1

Cn sinh(gn y/a) cos(gn x/a),

where the coefficients Cn remain to be identified and gn are the positive roots of
the transcendental equation g tan(g)=a h. Taking a=1, b=0.5, h=1, A=100,
solve the transcendental equation and plot the isotherms inside the bar.

04-S11: Locomotive on a Bridge
As a model of the motion of a locomotive across a railway bridge, consider a
periodically vibrating point load A sin(ω t) moving with constant velocity V
along a horizontal, uniform, rectangular steel beam of density ρ, cross-sectional
area S, and length L. The beam is fixed at x=0 and L and it is assumed that
at time t=0 the beam is at rest and the locomotive is at x=0. The equation
of motion for transverse vibrations (amplitude ψ) of the beam is given by

a4 ∂4ψ

∂t4
+

∂2ψ

∂t2
=

q(x, t)
ε

,

with q(x, t) = A sin(ω t) δ(x − V t). Here ε = ρ S is the linear density of the
beam and a = (κ2 Y/ρ)1/4 with κ the radius of gyration of the beam and
Y its Young’s modulus. Assuming a Fourier series of the form ψ(x, t) =∑∞

n=1 bn(t) sin(n π x/L), determine the transverse vibrations of the beam. Tak-
ing ρ=7800 kg/m3, Y =2.1 × 1011 N/m2, S =1 m2, κ=0.5 m, A=5 × 104 N,
ω = 1 s−1, L = 1000 m, and V = 1 m/s, animate the motion of the beam with
the motion of the locomotive along the beam superimposed. Keep 20 terms in
the series solution and use 100 frames in the animation. To see the oscillations,
unconstrained scaling should be used.

04-S12: The Temperature Switch
The temperature at the ends x = 0 and x = 100 of a rod (insulated on its
sides) 100 cm long is held at 0 ◦ and 100 ◦, respectively, until steady-state is
achieved. Then, at the instant t = 0, the temperature of the two ends is
interchanged. Determine the resultant temperature distribution T (x, t). If
a =

√
K/ρC = 10/π cm/s1/2, where K is the thermal conductivity of the rod,

ρ is its density, and C the specific heat, animate the temperature distribution.
What is the temperature at x = 20 cm after 25 seconds?

04-S13: Telegraph Equation Revisited
For an audio-frequency submarine cable [SR66], the telegraph equation applies
with the leakage constant G=0 and the self inductance (per unit length) L=0.
As shown in recipe 04-2-1, in this case the telegraph equation reduces to a
1-dimensional diffusion equation for the potential V with a diffusion constant
d = 1/(RC). Consider a submarine cable � = 1000 km in length, and let the
voltage at the source (at x = 0), under steady-state conditions be 1200 volts
and at the receiving end (at x = �) be 1100 volts. At time t = 0, the receiving
end is grounded, so that its voltage is reduced to zero, but the potential at the
source is maintained at its constant value of 1200 volts. If R=2 ohms/km and
C =3 × 10−7 farad/km, determine the current and voltage in the line after the
grounding of the receiving end and animate the results.

04-S14: Another “Trampoline” Example
A “trampoline” consists of a uniform, light, horizontal, stretched, rectangular
membrane with edges of length a and 2a. The edges at x=0, a are fixed, while
those at y=0, 2a are free. Using the separation of variables method, determine
the trampoline’s subsequent motion if it has the initial shape f = 2 x h/a for
0 ≤ x ≤ a/2 and f = 2 h (a − x)/a for a/2 ≤ x ≤ a and is released from rest.
Taking h=1/5 m, a=2 m, and c=2 m/s, animate the motion of the membrane
over the time interval t=0 to 5 seconds taking 100 frames.

04-S15: An Electrostatic Poisson Problem
The faces of a rectangular box, 0 ≤ x ≤ a, 0 ≤ y ≤ b, 0 ≤ z ≤ c are held at
the potential φ=0 and the interior is filled with charge with a charge density
ρ=A sin(π x/a) sin(π z/c) [y (y − b)] Coulombs/meter3. Using the electrostatic
Poisson equation ∇2φ=−ρ/ε0, where ε0 is the permittivity of free space, deter-
mine the potential at an arbitrary point inside the box. Taking A=10−9 C/m5,
a = b = c = 1 m, and ε0 = 8.85 × 10−12 Farads/meter, determine the potential
Φ (in volts) at the center of the box. Plot the equipotentials Φ/10, 2 Φ/10,...,
9 Φ/10 in the mid-plane x=a/2.

04-S16: SHE Does Not Want to Separate
Demonstrate that the scalar Helmholtz equation (SHE) does not separate in
bispherical coordinates even with the modified assumption that was successful
for Laplace’s equation in recipe 04-3-1.

04-S17: WE Can Separate
A 2-dimensional curvilinear coordinate system (u, v) can be defined through
the equations x=

√
u v, y=(u − v)/2, where the range of u, v is from 0 to ∞.

(a) Plot the contours in the x-y plane corresponding to holding u and v
constant. Show that this curvilinear coordinate system is orthogonal.

(b) Calculate the scale factors and the wave equation (WE) for this system.

(c) Show that WE is separable. Identify the separated ODEs and solutions.

04-S18: The Stark Effect
A hydrogenic atom consists of an electron of charge −e and mass m moving in
the attractive Coulomb field of a nucleus (atomic number Z) of charge Z e and
mass M . If Z =1, one has the hydrogen atom, Z =2 corresponds to the He+ ion,
Z =3 to the Li++ ion, and so on. The time-independent Schrödinger equation
for the wave function ψ then takes the form ∇2ψ + (2m/h̄2)[E + Z e2/r] ψ=0,
where h̄ is Planck’s constant divided by 2π, E is the total energy, and r the
radial distance of the electron from the nucleus. When a hydrogenic atom is
placed in an external electric field, the energy levels are found to shift. This
phenomenon is referred to as the Stark effect. If the electric field (magnitude
E0) is oriented in the positive z direction, a potential energy term −e E0 z must
be added to the hydrogenic atom problem. Show that the time-independent
Schrödinger equation is still separable in parabolic (or paraboloidal) coordinates
(ζ, η, φ) which are related to Cartesian coordinates by the relations x=ζ η cos φ,
y=ζ η sin φ, and z=(η2 − ζ2)/2, with 0 ≤ ζ < ∞, 0 ≤ η < ∞, 0 ≤ φ ≤ 2 π.

04-S19: Annular Temperature Distribution
An annular region, of inner radius r = 10 cm and outer radius 20 cm, has
its inner boundary maintained at the temperature (in degrees Celsius) T =
20 cos θ and the outer boundary held at T = 30 sin θ. Determine the steady-
state temperature distribution in the annular region and plot the isotherms
corresponding to −30, −20, −15,..., 0, 5, 10,..., 30 degrees.

04-S20: Split-boundary Temperature Problem
A thin circular plate of radius 1 m, whose two faces are insulated, has half of its
circular boundary kept at the constant temperature T1 and the other half at
the constant temperature T2. Find the steady-state temperature distribution
in the plate. Taking T1 = 300 degrees Celsius and T2 = 200 degrees Celsius,
plot the isotherms in 5 degree increments.

04-S21: Fluid Flow Around a Sphere
A solid sphere of radius a is placed in a fluid which was flowing uniformly with
speed V0 in the z direction. The velocity potential U for the fluid in the region
outside the sphere satisfies Laplace’s equation in spherical polar coordinates
and the velocity field is given by �v=−∇U . If the sphere is assumed to be rigid,
the normal component of �v must vanish at the surface of the sphere. Determine
the velocity field for the fluid outside the sphere and plot the velocity vectors.
Take a=1 m and V0 =1 m/s.

04-S22: Sound of Music?
Some musically inclined people like to sing in the shower stall when taking their
shower. In this problem, the shower stall is empty without the water running
and consists of a completely enclosed hollow vertical metal cylinder of radius a
and height h with (approximately) rigid walls. The speed of sound for the air
inside the cylinder is c. By solving the scalar Helmholtz equation for the spatial
part of the velocity potential, determine the allowed normal modes inside the
cylinder. For rigid walls, the normal component of the fluid velocity (or the

normal derivative of the potential) must vanish at each wall. Taking a = 1.83
m, h = 3.04 m, and c = 344 m/s, determine the three lowest eigenfrequencies.
By either consulting a musically inclined friend or a music reference book, or
going to the Internet, find the closest musical notes on the equal-tempered scale
to these eigenfrequencies.



Chapter 5

Complex Variables

5.1 Introduction

Let z = x + i y be a complex variable, with i ≡ √−1 and x and y real. For
given values of x and y, z is represented by a point in the complex z-plane (the
Argand plane) with x and y its coordinates on the real and imaginary axes,
respectively. In polar form, z can be written as z = r eiθ = r cos θ + i r sin θ,
with r=

√
x2 + y2 the radial distance (called the modulus) of z from the origin

and θ = arctan(y/x) the angle (called the argument) with the real axis . The
argument is not unique. If one writes it as θ + n(2 π), where 0 ≤ θ ≤ 2 π and
n== 0, ±1, ..., then θ is called the principal argument.

A complex function of z is of the general form ω=f(z)=u(x, y) + i v(x, y),
with f specified and u and v real. If, e.g., f(z) = z2, then ω = (x + i y)2 =
x2 − y2 + 2 i x y, so u=x2 − y2 and v=2 x y. f(z) is single-valued in a region R
if for each z in R there is only one value of ω. Otherwise, f(z) is multi-valued.
An example of the latter is f(z) = z1/2 = r1/2ei θ/2. For a single-valued f(z),
increasing θ from 0 to 2π yields the same value of the function. However, for
our example, f(z) = r1/2 for θ = 0, but f(z) = r1/2ei π = −r1/2 for θ = 2 π.
One must increase θ by 4π (2 revolutions around the origin) to regain +r1/2.
f(z)=z1/2 is said to have two branches and the origin is called a branch point.

f(z) is analytic at a point z if a unique derivative exists there, independent
of how z is approached. A necessary and sufficient condition for ω=f(z) to be
regular (single-valued and analytic) in R is that u and v satisfy the Cauchy–
Riemann (C-R) conditions,

∂u

∂x
=

∂v

∂y
,

∂v

∂x
= −∂u

∂y
, (5.1)

provided that the partial derivatives exist and are continuous in R. Assuming
the second derivatives exist, differentiating (5.1) with respect to x and y yields

∇2u =
∂2u

∂x2 +
∂2u

∂y2 = 0, ∇2v =
∂2v

∂x2 +
∂2v

∂y2 = 0, (5.2)

so u and v satisfy Laplace’s equation and thus represent possible real potentials.

5.1.1 Jennifer Tests Basics

There’s a basic rule which runs through all kinds of music, kind of
an unwritten rule. I don’t know what it is. But I’ve got it.
Ron Wood, British rock musician, Independent (London, 10 Sept. 1992)

As a test of the basic ideas presented in the introduction, Jennifer has pre-
pared the following quiz for her complex variables class:

Consider the single-valued, analytic, function
ω=u + i v=f(z)=z2ez2 − 5 cos3z, where z=x + i y.

(a) Determine u and v. Confirm that the Cauchy–Riemann conditions are
satisfied and that u and v satisfy Laplace’s equation.

(b) Form a new function F =f(z) z� = U + i V , where z� =x− i y. Show that
U and V do not satisfy the C-R conditions, so F is non-analytic.

Ms. Curious, who is currently in Jennifer’s class, has kindly supplied us with
her solution to this quiz. I will let her explain it to you in her own words.

“I will begin by assuming that x and y are real, to simplify later results, and
then enter the complex variable z = x + i y. On inputting the given complex
function f , z is automatically substituted.

> restart: assume(x::real,y::real):

> z:=x+I*y: f:=zˆ2*exp(zˆ2)-5*cos(z)ˆ3;

f := (x + y I)2 e((x+y I)2) − 5 cos(x + y I)3

Although I could grind out the forms of u and v by hand, it’s simpler to apply
the complex evaluation command to f , which splits it into real and imaginary
parts, and then extract u and v by taking the real and imaginary parts.

> f:=evalc(f); u:=Re(f); v:=Im(f);

u := (x2 − y2) e(x2−y2) cos(2 x y) − 2 x y e(x2−y2) sin(2x y)
− 5 cos(x)3 cosh(y)3 + 15 cos(x) cosh(y) sin(x)2 sinh(y)2

v := 2x y e(x2−y2) cos(2 x y) + (x2 − y2) e(x2−y2) sin(2x y)
+ 15 cos(x)2 cosh(y)2 sin(x) sinh(y) − 5 sin(x)3 sinh(y)3

I will check the C-R conditions using two different approaches. For the first,
∂u/∂x is calculated in ux and ∂v/∂y in vy (output suppressed). Forming ux−vy
in CR1 , and simplifying, yields 0, confirming the first C-R condition.

> ux:=diff(u,x); vy:=diff(v,y): CR1:=simplify(ux-vy);

ux := 2x e(x2−y2) cos(2 x y) + 2 (x2 − y2) x e(x2−y2) cos(2 x y)

− 2 (x2 − y2) e(x2−y2) sin(2x y) y − 2 y e(x2−y2) sin(2x y)

− 4 x2 y e(x2−y2) sin(2x y) − 4 x y2 e(x2−y2) cos(2 x y)
+ 15 cos(x)2 cosh(y)3 sin(x) − 15 sin(x)3 cosh(y) sinh(y)2

+ 30 cos(x)2 cosh(y) sin(x) sinh(y)2

CR1 := 0
For the second C-R condition, the “is” command is used to ask whether
∂v/∂x=−∂u/∂y, or not. The possible answers are either true or fail. As can
be seen in CR2 , the answer is true, thus confirming the second C-R condition.

> CR2:=is(diff(v,x)=-diff(u,y));

CR2 := true
To answer the last part of (a), I form an operator L which will calculate the
two-dimension Laplacian of an input function w and simplify the result.

> L:=w->simplify(diff(w,x,x)+diff(w,y,y));

L := w → simplify((
∂2

∂x2 w) + (
∂2

∂y2 w))

Applying L to u and v confirms that ∇2u=0 and ∇2v=0.

> Lu:=L(u); Lv:=L(v);

Lu := 0 Lv := 0
For part (b), the function F =f z� is formed and the real and imaginary parts
of F extracted in U and V , respectively.

> F:=f*conjugate(z): U:=Re(F); V:=Im(F);

U := e(x2−y2) cos(2 x y) x3 + x y2 e(x2−y2) cos(2 x y) − x2 y e(x2−y2) sin(2x y)

− e(x2−y2) sin(2x y) y3−5 x cos(x)3cosh(y)3+15 x cos(x)cosh(y) sin(x)2 sinh(y)2

+ 15 y cos(x)2 cosh(y)2 sin(x) sinh(y) − 5 y sin(x)3 sinh(y)3

V := e(x2−y2) cos(2 x y) y3 + x2 y e(x2−y2) cos(2 x y) + x y2 e(x2−y2) sin(2x y)

+ e(x2−y2) sin(2x y) x3+5 y cos(x)3cosh(y)3−15 y cos(x)cosh(y) sin(x)2 sinh(y)2

+ 15 x cos(x)2 cosh(y)2 sin(x) sinh(y) − 5 x sin(x)3 sinh(y)3

The combination ∂U/∂x − ∂V/∂y is calculated in CR1b and simplified. The
answer is non-zero for arbitrary (you can check it for specific values, if you
desire) values of x and y, so the first C-R condition isn’t satisfied.

> CR1b:=simplify(diff(U,x)-diff(V,y));

CR1b := 30 cos(x) cosh(y)3 − 30 cos(x) cosh(y) − 40 cos(x)3 cosh(y)3

+ 30 cos(x)3 cosh(y) − 4 x y e((x−y) (x+y)) sin(2x y)

+ 2 x2 e((x−y) (x+y)) cos(2 x y) − 2 y2 e((x−y) (x+y)) cos(2 x y)
The second C-R condition, calculated in CR2b, fails when the “is ” command
is applied.

> CR2b:=is(diff(V,x)=-diff(U,y));

CR2b := FAIL
The above failures imply that the single-valued function F is non-analytic.”

5.1.2 The Stream Function

Human kindness is like a defective tap, the first gush may be
impressive but the stream soon dries up.
P. D. James, British mystery writer, Devices and Desires, (1989)

In aerodynamics and fluid mechanics, the functions φ and ψ in the analytic
function f(z = x + i y)=φ(x, y) + i ψ(x, y) are called the velocity potential and
stream function, respectively. The velocity potential was introduced in Recipes
04-S21 and 04-S22. The curves ψ(x, y) = constant represent the tracks of the
fluid particles and are called streamlines. Consider φ = x2 + 4 x − y2 + 2 y.

(a) Confirm that φ satisfies Laplace’s equation so can represent the velocity
potential for steady-state fluid flow.

(b) Using the Cauchy–Riemann conditions, determine ψ(x, y).

(c) Make a contour plot, showing curves of constant φ and ψ. Use constrained
scaling to show that the two families of curves appear to be orthogonal.
Suggest a fluid flow problem where these contours might apply.

(d) Analytically show that the contours in (c) are orthogonal.

(e) Express f completely in terms of z.

The plots and VectorCalculus packages are loaded, the former needed for the
contourplot command, the latter for the Laplacian.

> restart: with(plots): with(VectorCalculus):

The given function φ(x, y) is entered.
> phi(x,y):=xˆ2+4*x-yˆ2+2*y;

φ(x, y) := x2 + 4 x − y2 + 2 y

Applying the Laplacian operator to φ in Cartesian coordinates yields 0, so φ
satisfies Laplace’s equation. Thus, φ is indeed a velocity potential for fluid flow.

> LE:=Laplacian(phi(x,y),’cartesian’[x,y]);

LE := 0
The first Cauchy–Riemann condition, ∂ψ/∂y=∂φ/∂x, is calculated in CR1 .

> CR1:=diff(psi(x,y),y)=diff(phi(x,y),x);

CR1 :=
∂

∂y
ψ(x, y) = 2x + 4

The form of ψ(x, y) is easily obtained by applying pdsolve to CR1 .
> sol1:=pdsolve(CR1,psi(x,y));

sol1 := ψ(x, y) = 2 y x + 4 y + F1(x)
The second C-R condition, ∂ψ/∂x=−∂φ/∂y, is calculated in CR2 .

> CR2:=diff(psi(x,y),x)=-diff(phi(x,y),y);

CR2 :=
∂

∂x
ψ(x, y) = 2 y − 2

An alternate form of ψ follows on applying pdsolve to CR2 .
> sol2:=pdsolve(CR2,psi(x,y));

sol2 := ψ(x, y) = 2 y x − 2 x + F1(y)
For the rhs of sol1 and sol2 to be the same, one must have F1(x)=−2x + C
and F1(y) = 4 y + C, where C is an arbitrary constant. These forms are
substituted into sol1 and sol2 .

> sol1b:=subs(_F1(x)=-2*x+C,sol1);

sol2b:=subs( F1(y)=4*y+C,sol2);

sol1b := ψ(x, y) = 2 y x + 4 y − 2 x + C

sol2b := ψ(x, y) = 2 y x + 4 y − 2 x + C

Taking the arbitrary constant C = 0 for plotting purposes, then ψ is given by
the right-hand side of sol1b (or sol2b).

> C:=0: psi(x,y):=rhs(sol1b);

ψ(x, y) := 2 y x + 4 y − 2 x

The contourplot command, with 19 contours and a 60 × 60 grid, is used to plot
φ(x, y) and ψ(x, y). The plot is colored by including the option filled=true
and boxed axes are chosen. The resulting picture is shown in Figure 5.1.

> contourplot({phi(x,y),psi(x,y)},x=-8..4,y=-5..7,
scaling=constrained,contours=19,grid=[60,60],
filled=true,axes=box,tickmarks=[3,3]);

–5

0

5

y

–5 0x

Figure 5.1: Streamlines and equipotentials.

The rectangular hyperboli in each quadrant are the streamlines and the curves
intersecting them at apparently 90◦ are the equipotentials. Considering, say,
the upper-right quarter of the figure, the streamlines could represent fluid flow
directions near the corner of two planar surfaces intersecting at right angles.

If, say φ(x, y) = constant, then dφ = (∂φ/∂x) dx + (∂φ/∂y) dy = 0, so the
slope of these curves is given by dy/dx=−(∂φ/∂x)/(∂φ/∂y). The slope of the

constant φ curves is calculated at an arbitrary point (x, y).
> slope[phi]:=-diff(phi(x,y),x)/diff(phi(x,y),y);

slopeφ := − 2 x + 4
−2 y + 2

The slope of the constant ψ curves is similarly determined.
> slope[Psi]:=-diff(psi(x,y),x)/diff(psi(x,y),y);

slopeΨ := −2 y − 2
2 x + 4

The product of the two slopes is calculated and found to be equal to −1, so the
two sets of curves are orthogonal.

> slope_product:=simplify(slope[phi]*slope[Psi]);

slope product := −1
The complex function f = φ(x, y) + i ψ(x, y) is entered, the forms of φ and ψ
being automatically entered. It is desired to express f entirely in terms of z.

> f:=phi(x,y)+I*psi(x,y);

f := x2 + 4 x − y2 + 2 y + (2 y x + 4 y − 2 x) I

Now, x=(z + z�)/2 and y=(z − z�)/(2i), where z� is the complex conjugate of
z. These forms are substituted into f . In the output, the symbol z represents
the conjugate of z.

> f:=subs({x=(z+conjugate(z))/2,y=(z-conjugate(z))/(2*I)},f);

f := (
z

2
+

1
2

z)2 + 2 z + 2 z +
1
4

(z − z)2 − (z − z) I

+ (−(z − z) (
z

2
+

1
2

z) I − 2 I (z − z) − z − z) I

On applying the simplify command, f is expressed completely in terms of z.
> f:=simplify(f);

f := z2 + 4 z − 2 I z

So, f(z)=z2 + 4 z − 2 i z, completing the solution of the problem.

5.2 Contour Integrals

If f(z) is regular within and on a simple closed curve C, the Cauchy–Riemann
conditions lead to Cauchy’s theorem,∮

C

f(z) dz = 0. (5.3)

Equivalently,
∫ z2

z1
f(z) dz has a value independent of the path joining two points

z1 and z2. If z0 is any point inside C, Cauchy’s theorem may be used in turn
to derive Cauchy’s integral formula,∮

C

f(z)
(z − z0)

dz = 2π i f(z0), (5.4)

where C is traversed in a counter-clockwise direction. Note that the integrand
is not analytic at z =z0. This point is called a first order pole or a simple pole
of the integrand. It is an example of an isolated singularity.

Cauchy’s formula may be differentiated n − 1 times to yield∮
C

f(z)
(z − z0)n

dz =
2 π i

(n − 1)!

(
dn−1f(z)

dzn−1

)
z = z0

, (5.5)

or, on setting g(z) ≡ f(z)/(z − z0)n,∮
C

g(z) dz =
2 π i

(n − 1)!

(
dn−1[(z − z0)ng(z)]

dzn−1

)
z = z0

. (5.6)

The isolated singularity in g(z) is called an nth order pole, while the coefficient
of 2π i on the rhs of (5.6) is referred to as the residue of g(z) at z=z0.

The above formulas may be used to prove Cauchy’s residue theorem: If g(z)
is regular within and on the closed contour C, except for a finite number of
poles, then

∮
g(z) dz = 2πi ×(Sum of the residues of g(z) at its poles within C).

5.2.1 Jennifer Tests Cauchy’s Theorem

The test of a real comedian is whether you laugh at him
before he opens his mouth.
George Jean Nathan, American critic, (1882–1958)

As a follow up to her earlier quiz, Jennifer has asked her complex variables class
to confirm that Cauchy’s theorem is satisfied for the regular complex function
f(z=x + i y=r eiθ)=z2ez2 − 5 cos3z for the following two closed contours C:

(a) C1: (i) along the x-axis from the origin to x = 1, (ii) vertically upwards
along x = 1 to y = 1, (iii) along y = 1 from x = 1 to x = 0, (iv) vertically
downwards along x=0 from y=1 back to the origin;

(b) C2: (i) radially outwards along the x-axis from the origin to radius r=R,
(ii) along a circular arc of radius R from the x-axis to the y-axis,
(iii) radially inwards along the y-axis from R to the origin.

Again, Ms. Curious has been asked to present and discuss the recipe which she
has created to solve this problem.

“In (a), C1 is such that rectangular coordinates should be used. Letting
f = u + i v, the general line integral is

∫
f(z) dz =

∫
(u + i v) (dx + i dy) =∫

(u dx − v dy) + i
∫

(v dx + u dy). Assuming that x and y are real, f is entered
along with z=x + i y.”

> restart: assume(x::real,y::real):

> f:=zˆ2*exp(zˆ2)-5*cos(z)ˆ3; z:=x+I*y:

f := z2 e(z2) − 5 cos(z)3

“The real and imaginary parts of f are determined (output suppressed here).”
> u:=Re(f); v:=Im(f);

“For the first leg, the integral is I1 =
∫ 1
0 (u+i v) dx, with the integrand evaluated

at y=0. The answer is expressed in terms of the error function (erf).”
> I1:=int(eval(u+I*v,y=0),x=0..1);

I1 :=
1
2

e +
1
4

I
√

π erf(I) − 5
3

cos(1)2 sin(1) − 10
3

sin(1)

“For the second leg, the integral is I2 =
∫ 1
0 (−v + i u) dy, with x=1.”

> I2:= int(eval(-v+I*u,x=1),y=0..1);

I2 := − 1
24

(−90 sin(1) e(3 I) − 10 sin(3) e(3 I) + 12 e(1+3 I)

+ 6 I
√

π erf(I) e(3 I) − 12 I e(5 I) − 12 e(5 I) + 6 I
√

π erf(1 − I) e(3 I)

+ 5 I e3 + 45 I e(1+2 I) − 5 I e(−3+6 I) − 45 I e(−1+4 I))e(−3 I)

“For the third leg, the integral is I3 =
∫ 0
1 (u + i v) dx, with y=1.”

> I3:= int(eval(u+I*v,y=1),x=1..0);

I3 :=
1
24

(5 I e(−3+3 I) − 45 I e(1+3 I) + 57 I e(−1+3 I) − 6 I
√

π erf(1) e(3 I)

− 5 I e(3+3 I) − 12 I e(5 I) − 12 e(5 I) + 6 I
√

π erf(1 − I) e(3 I)

+ 5 I e3 + 45 I e(1+2 I) − 5 I e(−3+6 I) − 45 I e(−1+4 I))e(−3 I)

“For the fourth leg, the integral is I4 =
∫ 0
1 (−v + i u) dy, with x=0.”

> I4:=int(eval(-v+I*u,x=0),y=1..0);

I4 := − 5
24

I e(−3) +
15
8

I e − 19
8

I e(−1) +
1
4

I
√

π erf(1) +
5
24

I e3

“Adding the four integrals and simplifying the complete contour integral CI ”
> CI:=simplify(I1+I2+I3+I4);

CI := 0
“yields 0, thus confirming Cauchy’s theorem. To deal with part (b), I will
unassign the Cartesian form of z, by enclosing z in right quotes, and enter its
polar form, z=reiθ. The complex function f then is as follows.”

> z:=’z’: z:=r*exp(I*theta): f:=f;

f := r2 (e(θ I))2 e(r2 (e(θ I))2) − 5 cos(r e(θ I))3

“For the first leg of the new contour C2, f is evaluated at θ=0 and integrated
from r=0 to R.”

> I1b:=int(eval(f,theta=0),r=0..R);

I1b :=
1
2

e(R2) R +
1
4

I
√

π erf(R I) − 5
3

cos(R)2 sin(R) − 10
3

sin(R)

“For the second leg, the integral is of the form
∫

f(z) dz=
∫ π/2
0 f(Reiθ) i Reiθ dθ.

This integration is carried out in I2b.”

> I2b:=int(eval(f*I*z,r=R),theta=0..Pi/2);

I2b :=
1
12

(−6 e(2 R2) R − 3 I
√

π erf(R I) e(R2) + 20 cos(R)2 sin(R) e(R2)

+ 40 sin(R) e(R2) + 6 I R − 3 I
√

π erf(R) e(R2)

− 20 I cosh(R)2 sinh(R) e(R2) − 40 I sinh(R) e(R2))e(−R2)

“The third leg is radially inwards along the y axis, for which θ = π/2. The
relevant integral is performed in I3b.”

> I3b:=int(eval(f*z/r,theta=Pi/2),r=R..0);

I3b :=
−1
12

I(6 R − 3
√

π erf(R) e(R2) − 20 cosh(R)2 sinh(R) e(R2)

− 40 sinh(R) e(R2))e(−R2)

“Adding the three integrals and simplifying, the complete contour integral CIb”
> CIb:=simplify(I1b+I2b+I3b);

CIb := 0
“is zero, once again confirming Cauchy’s theorem.”

5.2.2 Cauchy’s Residue Theorem

To know yet to think that one does not know is best;
Not to know yet to think that one knows will lead to difficulty.
Lao–Tzu, Chinese philosopher, 6th century BC

As a follow-up quiz to that posed in the last recipe, here’s one provided by
Jennifer for which the integrands of the contour integrals contain isolated sin-
gularities, resulting in non-zero values for the integrals. The contours are to be
traversed in a counterclockwise sense

(a) Evaluate
∮

C
(5 z4 − 3 z2 + 2)/(z − 1)n, with n = 2, 3, 4, 5, where C is any

simple closed curve enclosing z=1. Identify the types of singularities.

(b) Evaluate
∮

C
(2 z3 + z)/((z2 − 1/4)(z2 + 2 z + 2)), where the contour C is

given by (i) |z|=3/2, (ii) |z|=5/8. Identify the singular points and make
a plot showing their locations in the z-plane and the two contours.

The plottools library package is needed to plot the contours in part (b).
> restart: with(plots): with(plottools):

Now, let’s tackle part (a). An operator g is introduced for generating the given
integrand (5 z4 −3 z2 +2)/(z −1)n for different input values of n. Clearly, since
the numerator remains finite there, the integrand has second, third, fourth and
fifth order poles at z=1 for n=2, 3, 4, 5, respectively.

> g:=n->(5*zˆ4-3*zˆ2+2)/(z-1)ˆn;

g := n → 5 z4 − 3 z2 + 2
(z − 1)n

A second operator F is formed for evaluating the integral for the nth order pole
using the Cauchy integral result in Equation (5.5).

> F:=n->2*Pi*I*eval(diff(numer(g(n)),z$(n-1)),z=1)/(n-1)!;

F := n →
2 I π (

dn−1

dzn−1 numer(g(n))) z = 1
(n − 1)!

Making use of F , the contour integral is evaluated for n = 2, ..., 5, the results
being displayed in II(2), ..., II(5).

> seq(II(n)=F(n),n=2..5);

II(2) = 28 I π, II(3) = 54 I π, II(4) = 40 I π, II(5) = 10 I π

An even easier approach to evaluating the contour integrals is to use the residue
command. An operator G is created to evaluate the residue of g(n) at z = 1
and multiply the result by 2π i.

> G:=n->2*Pi*I*residue(g(n),z=1);

G := n → 2 I π residue(g(n), z = 1)
As a check, G(n) is calculated for n=2 to 5, the answers given for II(2) to II(5)
being the same as obtained above.

> check:=seq(II(n)=G(n),n=2..5);

check := II(2) = 28 I π, II(3) = 54 I π, II(4) = 40 I π, II(5) = 10 I π

The integrand for part (b) is now entered in g2 .
> g2:=(2*zˆ3+z)/((zˆ2-1/4)*(zˆ2+2*z+2));

g2 :=
2 z3 + z

(z2 − 1
4
) (z2 + 2 z + 2)

One approach to evaluating the contour integral is to first convert g2 to a
partial fraction form in terms of z. The complex option allows a complete
decomposition (in floating point form) in terms of the complex roots.

> g2b:=convert(g2,parfrac,z,complex);

g2b :=
0.5846153848 − 1.323076923 I

z + 1. + 1.000000000 I
+

0.6000000002
z + 0.5000000000

+
0.5846153848 + 1.323076923 I

z + 1. − 1.000000000 I
+

0.2307692307
z − 0.5000000000

g2b can be converted to a somewhat more compact rational form.
> g2c:=convert(g2b,rational);

g2c :=

38
65

− 86
65

I

z + 1 + I
+

3

5 (z +
1
2
)

+

38
65

+
86
65

I

z + 1 − I
+

3

13 (z − 1
2
)

Examining g2c, the integrand g2 has four first order (simple) poles located at
z = 1/2, −1/2, −1 + i and −1 − i. Using the pointplot command, the four
simple poles are plotted as size 16 blue circles in the complex z-plane in pp.

> pp:=pointplot([[1/2,0],[-1/2,0],[-1,1],[-1,-1]],

symbol=circle,symbolsize=16,color=blue):
The specified contours |z|=5/8 and |z|=3/2 are circles centered on the origin
of radii 5/8 and 3/2, respectively. These circles are created in c1 and c2, the
smaller circle being colored red, the larger one colored green.

> c1:=circle([0,0],5/8,color=red,thickness=2):

> c2:=circle([0,0],3/2,color=green,thickness=2):

The three graphs, pp, c1, and c2, are superimposed with the display command,
the resulting picture being shown in Figure 5.2. The larger circular contour
encloses all four poles, the smaller circle enclosing only the poles at z=±1/2.

> display({pp,c1,c2},labels=["x","y"]);

–1

1
y

–1 1
x

Figure 5.2: Four simple poles of g2 and two circular contours.

Cauchy’s residue theorem states that the contour integral is equal to 2πi times
the sum of the residues of the poles enclosed by the contour. Let’s first consider
the contour |z| = 3/2, which encloses all 4 poles. In g2c, this means that one
simply has to add the numerators of all 4 terms. The numerator of the ith term
in g2c can be extracted with the operand command op[i,1]. The 4 numerators
are then added and multiplied by2πi, yielding the answer 4πi for the integral.

> I2:=2*Pi*I*add(op([i,1],g2c),i=1..4);

I2 := 4 I π

As shown in check2 , the same answer follows on applying the residue command
directly to g2 at each pole, adding the four residues, and multiplying by 2πi.

> check2:=2*Pi*I*(residue(g2,z=1/2)+residue(g2,z=-1/2)

+residue(g2,z=-1+I)+residue(g2,z=-1-I));

check2 := 4 I π

Using the residue command, the value of the contour integral is now obtained
for the contour |z|=5/8 by only keeping the residues of the poles at z=±1/2.

> I2b:=2*Pi*I*(residue(g2,z=1/2)+residue(g2,z=-1/2));

I2b :=
108
65

I π

In this latter case, the contour integral has the value (108/65)πi.

5.3 Definite Integrals

By performing a closed contour integration in the complex z-plane with a suit-
ably chosen path and using Cauchy’s residue theorem, it is possible to easily
evaluate some real definite integrals. The choice of path depends on the form
of the integral. A few representative examples will now be considered.

5.3.1 Infinite Limits

God does not care about our mathematical difficulties.
He integrates empirically.
Albert Einstein, Nobel laureate in physics, (1879–1955)

To evaluate an integral of the form I =
∫∞

−∞ f(x) dx using contour integration,
a “standard” approach is to evaluate J =

∮
C

f(z) dz with C a closed semi-
circle of radius R in the complex z-plane with its flat portion along the real
axis. Along the real axis, z =x and the contribution to J is J1 =

∫ R

−R
f(x) dx.

In the limit as R → ∞, J1 → I, the original integral whose value we seek.
Along the semi-circular arc, z =Reiθ and the line integral contribution to J is
J2 =

∫ π

0 f(R eiθ) iReiθ dθ, if the arc is taken in the upper-half z-plane. If J2 → 0
as R → ∞, evaluating J will be equivalent to evaluating I. The value of J
is then determined by calculating the sum of the residues of its poles inside C
and multiplying by 2πi. As an illustrative example, let’s evaluate the integral

I =
∫ ∞

−∞
dx/(1 + x4).

As suggested, we consider the contour integral J =
∮

C
dz/(1 + z4) with C a

closed semi-circle of radius R in the upper-half z-plane and its flat portion along
the real x-axis. The plottools library package is loaded so that the semi-circular
arc can be plotted, and the integrand of J is then entered.

> restart: with(plots): with(plottools):

> integrand:=1/(1+zˆ4);

integrand :=
1

1 + z4

Along the semi-circle, the line integral is J2 =
∫ π

0 iReiθ dθ/(1 + (Reiθ)4). For
large R, the integrand of J2 has a 1/R3 dependence, so J2 → 0 as R → ∞.
The real axis contribution becomes the original integral. To find its value, we
must determine the sum of the residues of any poles inside C. To this end, the
denominator of the integrand is set equal to zero and solved for the z roots.

> Z:=solve(denom(integrand)=0,z);

Z :=
√

2
2

+
1
2

I
√

2,
1
2

I
√

2 −
√

2
2

, −
√

2
2

− 1
2

I
√

2, −1
2

I
√

2 +
√

2
2

There are four complex roots which locate four simple poles in the complex
z-plane. To see which poles contribute to the contour integral, a semi-circular
path of radius R=2 will be drawn in the upper-half z-plane.

> R:=2:

The arc command is used to draw a red semi-circle of radius R centered on the
origin (0,0). In the command, the angle is allowed to vary from 0 to π radians.

> a:=arc([0,0],R,0..Pi,color=red,thickness=2):

The flat portion of the contour between (−R, 0) and (R, 0) is plotted by entering
the end points as a list of lists and choosing a line style.

> b:=plot([[-R,0],[R,0]],style=line,color=red,thickness=2):

Taking the real and imaginary parts of each root and putting them into a list
of lists, the locations of the four poles are plotted as size 16 blue circles.

> c:=pointplot([seq([Re(Z[i]),Im(Z[i])],i=1..4)],

symbol=circle,symbolsize=16,color=blue):
The three graphs are superimposed with constrained scaling, the resulting pic-
ture being shown in Figure 5.3.

> display({a,b,c},scaling=constrained,view=[-R..R,-R..R]);

–2

–1

0

1

2

–2 –1 1 2

Figure 5.3: Locations of the four poles and the semi-circular contour.

Two of the poles lie inside the contour, the remaining two outside. Only the
former poles contribute to the closed contour integral. The sum r of their
residues is now calculated.

> r:=residue(integrand,z=Z[1])+residue(integrand,z=Z[2]);

r :=
1

2 I
√

2 − 2
√

2
+

1
2 I

√
2 + 2

√
2

The value of the integral follows on multiplying r by 2π i and applying the
complex evaluation command to simplify the answer to a real number.

> V:=evalc(2*Pi*I*r);

V :=
π

√
2

2
As a check, we can directly evaluate the integral using Maple’s int command.
This is done in the next two command lines, the resulting number being the
same as that obtained above using contour integration.

> I1:=Int(integrand,z=-infinity..infinity);

I1 :=
∫ ∞

−∞

1
1 + z4 dz

> check:=value(I1);

check :=
π

√
2

2
If the semi-circular contour had been chosen to be in the lower-half z-plane,
exactly the same answer would be generated using the poles below the real
axis, provided that the result is multiplied by −1 to take into account that the
contour then is taken in a clockwise sense, rather than counterclockwise.

5.3.2 Poles on the Contour

There are many paths to the top of the mountain,
but the view is always the same.
Chinese Proverb

In the previous example, the poles of the integrand were either definitely inside
or outside the path chosen for the contour integration. Suppose that we want

to evaluate the integral I =
∫ ∞

−∞
cos(x) dx/(a2 − x2) with a real. The infinite

limits suggest a semi-circular contour C in the complex z-plane with the flat
portion on the real (x) axis. But the integrand has two simple poles lying right
on C at x = ±a. To apply Cauchy’s residue theorem, the poles must be def-
initely inside or outside the chosen contour. So what do we do? A standard
approach is to “indent” the contour in the vicinity of each pole, taking a small
semi-circular “detour” around the pole which either puts the pole inside or out-
side the contour. Then one performs the contour integration using the residue
theorem, and afterwards lets the radii of the small semi-circles go to zero. As
an illustration of this method, let’s evaluate the above integral I.

The plottools library package is again loaded so the various semi-circles can
be drawn for the indented contour.

> restart: with(plots): with(plottools):

Instead of tackling I directly, let’s consider the integral
∮

C
eizdz/(a2 −z2), with

z the complex variable. At the end of the calculation the real integral I can be

obtained by taking the real and imaginary parts. The integrand of the complex
closed contour integral is now entered.

> integrand:=exp(I*z)/(aˆ2-zˆ2);

integrand :=
e(z I)

a2 − z2

The integrand’s poles can be extracted by applying the discontinuity command.
> pole:=discont(integrand,z);

pole := {a, −a}
To draw the indented semi-circular contour, let’s take the poles to be at ±1 by
setting a=1. The small semi-circle around each pole is given a radius r =0.2,
and the larger semi-circle which completes the path is given a radius R=2.

> R:=2: r:=0.2: a:=1:

An operator A is formed to draw a thick red semi-circle of arbitrary radius r
around a specified point (x, 0) on the real axis.

> A:=(x,r)->arc([x,0],r,0..Pi,color=red,thickness=2):

Using A, a large semi-circle of radius R centered on the origin is produced in
A1, and a small semi-circle of radius r around each pole in A2 and A3.

> A1:=A(0,R): A2:=A(pole[1],r): A3:=A(pole[2],r):

In B, a thick red line is drawn along the real axis between, x=−R and −a − r,
between −a + r and a − r, and between a + r and R.

> B:=plot([[[-R,0],[-a-r,0]],[[-a+r,0],[a-r,0]],

[[a+r,0],[R,0]]],style=line,color=red,thickness=2):
In C, the two poles are plotted as size 16 blue circles.

> C:=pointplot([[pole[1],0],[pole[2],0]],symbol=circle,

symbolsize=16,color=blue):
The graphs are superimposed, generating the full contour shown in Figure 5.4.

> display({A1,A2,A3,B,C},scaling=constrained,
view=[-R..R,0..R],tickmarks=[2,2],labels=["x","y"]);

0

1

2

y

–2 2x

Figure 5.4: Semi-circular contour indented around the two poles on the x-axis.

Note that I have chosen to indent the small semi-circles around each pole in
such a way that both poles lie outside the contour C. So the line integral around
C will yield zero, since no poles lie inside C. I could just as well have chosen
to indent the small semi-circles in such a way that both poles were inside the
contour, or even so that one pole was inside and the other outside. As you may
verify, the final answer is the same, irrespective of the choice of indentation.

The quantities a and r are now unassigned. On each small semi-circle, the
complex variable will be written as Z =reiθ.

> a:=’a’: r:=’r’: Z:=r*exp(I*theta):

For our choice of C, starting at the point (−R, 0) the complete counterclockwise
contour integral will be as follows:

∮
C

eizdz

(a2 − z2)
=0 =

∫ −a−r

−R

eixdx

(a2 − x2)
+
∫ 0

π

ei(−a+Z) iZ dθ

(a2 − (−a + Z)2)
+
∫ a−r

−a+r

eixdx

(a2 − x2)

+
∫ 0

π

ei(a+Z) iZ dθ

(a2 − (a + Z)2)
+
∫ R

a+r

eixdx

(a2 − x2)
+
∫ π

0

eiReiθ

iReiθ dθ

(a2 − (Reiθ)2)
.

In the limit that R → ∞ and r → 0, the sum of the first, third, and fifth
integrals on the right-hand side of 0 will yield

∫∞
−∞ eix dx/(a2 − x2). Taking

the real part will produce the original integral
∫∞

−∞ cos(x) dx/(a2 − x2), whose
value we are trying to determine.

Let’s look at the last integral on the right-hand side, which is written as∫ π

0

eiR cos θ−R sin θ iReiθ dθ

(a2 − R2e2iθ)
.

Since the large semi-circle is in the upper-half z-plane, sin θ > 0. Therefore,
as R → ∞, the integrand → 0 and the integral contribution is zero. Finally,
we have to evaluate the second and fourth integrals on the right-hand side. A
function F is formed for substituting z=±a + Z into the integrand.

> F:=n->subs(z=pole[n]+Z,integrand):

The first pole extracted was +a (the ordering varies from one run to the next),
so F(1) is used to form the fourth integral and simplify it.

> I1:=simplify(Int(F(1)*I*Z,theta=Pi..0));

I1 := −I

∫ 0

π

e((a+r e(θ I)) I)

2 a + r e(θ I) dθ

Taking the limit of I1 at r=0 completes the evaluation of the integral in I1b.
> I1b:=limit(I1,r=0);

I1b :=

1
2

I e(a I) π

a
Using F(2), the second integral is similarly formed in I2 and evaluated in I2b.

> I2:=simplify(Int(F(2)*I*Z,theta=Pi..0));

I2 := −I

∫ 0

π

e((−a+r e(θ I)) I)

−2 a + r e(θ I) dθ

> I2b:=limit(I2,r=0);

I2b :=

−1
2

I e(−I a) π

a
The two integral terms are added and simplified assuming that a is real.

> terms:=simplify(I1b+I2b) assuming a::real;

terms := −π sin(a)
a

The answer to the original real integral is just minus the above output.
> answer:=Int(cos(x)/(aˆ2-xˆ2),x=-infinity..infinity)=-terms;

answer :=
∫ ∞

−∞

cos(x)
a2 − x2 dx =

π sin(a)
a

5.3.3 An Angular Integral

Science is an integral part of culture. ... It’s one of the glories of
the human intellectual tradition.
Stephen Jay Gould, American scientist, (b. 1941)

Integrals whose integrands are rational functions of cos θ and sin θ in the range
of integration can be evaluated by making the transformation z=eiθ. If θ varies
from 0 to 2π, the integration in the complex z-plane will be counter-clockwise
around a circle of unit radius. As an example, consider

J =
∫ 2π

0

sin2 θ dθ

(5 + 4 cos θ)
.

For variety, let’s load the Student[Calculus1] package, which contains two useful
commands, Integrand and Roots, which haven’t been used yet.

> restart: with(Student[Calculus1]):

The integral J is now entered and, as a check on the contour integration, eval-
uated directly with the value command.

> J:=Int(sin(theta)ˆ2/(5+4*cos(theta)),theta=0..2*Pi);

J :=
∫ 2 π

0

sin(θ)2

5 + 4 cos(θ)
dθ

> check:=value(J);

check :=
π

4
A contour integration should yield the value π/4 for J . To carry out this
integration, note that sin θ = (eiθ − e−iθ)/(2i) = (z − 1/z)/(2i) and cos θ =
(eiθ + e−iθ)/2=(z + 1/z)/2. These two transformations are entered.

> sin(theta):=(z-1/z)/(2*I); cos(theta):=(z+1/z)/2;

sin(θ) :=
−1
2

I (z − 1
z
) cos(θ) :=

z

2
+

1
2 z

Using the Integrand command to extract the integrand of J and using the fact
that dθ=dz/(iz), the integrand of the contour integral is as follows.

> integrand:=simplify(Integrand(J)/(I*z));

integrand :=

1
4

I (z2 − 1)2

z2 (5 z + 2 z2 + 2)
The Roots command is used to determine the zeros of the reciprocal of the
integrand, and thus the poles of the integrand.

> R:=Roots(1/integrand);

R := [−2,
−1
2

, 0]

There are three poles, viz., two simple poles at z = −2 and z = −1/2 and a
second order pole at z =0. Since the contour C is a unit circle in the z-plane,
the pole at z =−2 will not contribute to the answer since it’s outside C. The
residue of the integrand at the two poles inside C is calculated in r1 and r2 .

> r1:=residue(integrand,z=R[2]); r2:=residue(integrand,z=R[3]);

r1 :=
3
16

I r2 :=
−5
16

I

By Cauchy’s residue theorem, the value of the integral J is just 2π i times the
sum of the two residues. The answer is π/4 as expected.

> answer:=2*Pi*I*(r1+r2);

answer :=
π

4

5.3.4 A Branch Cut

Trust everybody, but cut the cards.
Finley Peter Dunne, American journalist, (1867–1936)

Cauchy’s residue theorem is built on the assumption that the integrand f(z)
in the contour integral

∮
C

f(z) dz is a single-valued function. What if f(z) is
multi-valued with more than one mathematical branch? The answer is to choose
a contour which keeps f(z) on a single branch. Typically, a line is drawn in
the complex z-plane, called a branch cut, which the contour is not permitted to
cross in order to keep f(z) single-valued. Section 5.5 will elaborate on this.

To see how the procedure is implemented for a specific integral, consider

J =
∫ ∞

0

√
x dx

1 + x2 . The corresponding complex integral is
∮

C
(
√

z/(1 + z2)) dz,

which has a double-valued integrand because of the
√

z factor. The branch cut
will be taken from z = 0 (the branch point) to −∞ along the negative x-axis.
(The branch cut could be taken in any direction from z=0 to ∞.) The contour

C is chosen to consist of an inner circle of radius ε and an outer circle of radius
R (both centered on the origin) joined by horizontal lines a distance η above
and below the branch cut. Ultimately, we let both ε and η → 0 and R → ∞.

In order to draw the circular arcs in C, the plottools library package is first
loaded. The complex integrand is then entered.

> restart: with(plots): with(plottools):

> integrand:=sqrt(z)/(1+zˆ2);

integrand :=
√

z

1 + z2

Before drawing C, let’s use the singular command to extract the singularities
of the integrand, needed for the application of Cauchy’s residue theorem.

> pole:=singular(integrand);

pole := {z = I}, {z = −I}
There are two singular points at z= i and z=−i, corresponding to the locations
of two simple poles. The operand command, op, is now used to separately obtain
the locations, z1 and z2 , of the two poles.

> z1:=op([1,2],pole[1]); z2:=op([1,2],pole[2]);

z1 := I z2 := −I

The pole locations may be converted into polar form.
> z1_polar:=convert(z1,polar); z2_polar:=convert(z2,polar);

z1 polar := polar(1,
π

2
) z2 polar := polar(1, −π

2
)

The poles lie on a circle of radius 1 at the polar angles θ=+π/2 and −π/2. To
schematically draw the contour, let’s set R=2, ε=0.2, and η=0.05.

> R:=2: epsilon:=0.2: eta:=0.05:

Thick red circular arcs of radius ε and R, respectively, are drawn in A1 and A2.
> A1:=arc([0,0],epsilon,Pi-0.2..-Pi+0.2,color=red,thickness=2):

> A2:=arc([0,0],R,Pi-0.02..-Pi+0.02,color=red,thickness=2):

A thick blue line is drawn along the real axis between x = −R − 0.5 and the
origin to represent the branch cut.

> B:=plot([[-R-0.5,0],[0,0]],style=line,color=blue,thickness=3):

Red lines are drawn between (−R,η), (−ε,η) and between (−R,−η), (−ε,−η).
> B2:=plot([[[-R,eta],[-epsilon,eta]],[[-R,-eta],

[-epsilon,-eta]]],style=line,color=red,thickness=2):
The two poles are represented by size 16 blue circles,

> C:=plot({[0,z1/I],[0,z2/I]},style=point,symbol=circle,
symbolsize=16,color=blue):

and the textplot command is used to add labels to the figure.
> tp:=textplot([[-1.9,-.15,"a"],[-1.9,.15,"b"],[-.25,.15,"c"],

[-.25,-.15,"d"],[-2.3,.1,"cut"]]):
The six plots are superimposed, the resulting picture being shown in Figure 5.5.

> display({A1,A2,B,B2,C,tp},scaling=constrained,
view=[-R-0.5..R,-R..R],labels=["x","y"]);

cut
d

cb
a

–1

1

y

–1 1 x

Figure 5.5: Closed contour around branch cut (stretching from z=0 to −∞).

The complete contour C consists of four legs, viz., a → b, b → c, c → d, and
d → a. Traversing C counter-clockwise, in the limit that η → 0, the increase of
2 π in θ in going from a to b is canceled out by the −2 π contribution in going
from c to d. Since the net angular change is zero, f(z) remains single-valued.
Multi-valuedness of f(z)=

√
z/(1+z2) only occurs if the net change in θ exceeds

2 π. Since f(z) is single-valued, Cauchy’s residue theorem can be applied. The
poles lie inside C, so the line integral around C will yield a non-zero result.
Let’s calculate the residues of the integrand at the two poles.

> r1:=residue(integrand,z=z1); r2:=residue(integrand,z=z2);

r1 :=
−1
2

I (
√

2
2

+
1
2

I
√

2) r2 :=
1
2

I (
√

2
2

− 1
2

I
√

2)

Now let’s look at each contribution to the line integral in the limit that η → 0.

(a) The integral Ia→b =
∫ π

−π
(i R3/2e3iθ/2/(1 + R2e2iθ)) dθ → 0 as R → ∞.

(b) Ib→c =
∫ ε

R
(r1/2 e3πi/2/(1 + r2e2πi)) dr → i J as ε → 0 and R → ∞.

(c) Ic→d =
∫ −π

π
(i ε3/2e3iθ/2/(1 + ε2e2iθ)) dθ → 0 as ε → 0.

(d) Id→a =
∫ R

ε
(r1/2 e−3πi/2/(1 + r2e−2πi)) dr → i J as ε → 0 and R → ∞.

Since these four contributions add up to 2 i J , the original integral must equal
2 π i times the sum of the two residues divided by 2 i.

> answer:=Int(sqrt(x)/(1+xˆ2),x=0..infinity)

=evalc(2*Pi*I*(r1+r2)/(2*I));

answer :=
∫ ∞

0

√
x

1 + x2 dx =
π

√
2

2
As a check, we apply the value command to the left-hand side of answer ,

> check:=value(lhs(answer));

check :=
π

√
2

2
obtaining exactly the same result as in the contour integration.

5.4 Laurent Expansion

Using Cauchy’s integral formula, one can prove Laurent’s theorem: If g(z) is
regular in an annular region R between two concentric circles with center z0,
then g(z) may be represented in R by a Laurent expansion

g(z) =
∞∑

n=−∞
an (z − z0)n, with an =

1
2 π i

∮
C

g(z) dz

(z − z0)n+1 , (5.7)

C being any simple closed path encircling z0 counter-clockwise within R.
If g(z) has a finite number of terms with negative exponents, it’s behavior is

dominated as z approaches z0 by the term with the largest negative exponent.
If, e.g., it has the form

g(z) =
a−n

(z − z0)n
+

a−(n−1)

(z − z0)n−1 + · · · + a−1

(z − z0)
+ a0 + a1(z − z0) + · · · , (5.8)

then it has a pole of order n at z = z0. Note that if g(z) in (5.8) is multiplied
by (z − z0)n, the result then differentiated n − 1 times and z set equal to z0,
then the coefficient a−1 is just the residue of the nth order pole, i.e.,

a−1 =
1

(n − 1)!

(
dn−1[(z − z0)ng(z)]

dzn−1

)
z = z0

. (5.9)

If g(z) has an infinite number of terms with negative exponents, it is said to

have an essential singularity at z=z0. For example, g(z)=e
1
z =1+

1
z

+
1
z2 + · · ·

has an essential singularity at z=0.
Instead of doing the contour integrations in (5.7), the coefficients an can

often be more easily obtained by considering the Taylor series of (z − z0)ng(z).

5.4.1 Ms. Curious Meets Mr. Laurent

Isn’t life a series of images that change as they repeat themselves?
Andy Warhol, American pop artist, (1928–87)

Jennifer has posed the following question for her complex variables class:
Keeping 10 terms, determine the Laurent expansion of f =cos(z) ez−2z2

/z2

about z =0. Identify the singularity and give the region of convergence of the
series. Determine the residue of f at the singularity from the series. Confirm
the value of the residue by alternate means.

Here is the solution presented by Ms. Curious, expressed in her own words.

“The number N of terms to be retained in the series is set equal to 10.

> restart: N:=10:

Since the numerator remains finite there, clearly f has a second-order pole at
z = 0. To obtain the Laurent series about this singularity, I will enter the
coefficient of 1/z2 in G.

> G:=cos(z)*exp(z-2*zˆ2);

G := cos(z) e(z−2 z2)

Then G is Taylor expanded in powers of z to order N , and the order of term
removed with the convert(g,polynom) command.

> g:=taylor(G,z,N);

> g:=convert(g,polynom);

g := 1+z − 2 z2 − 7
3

z3+
11
6

z4+
79
30

z5 − z6 − 1217
630

z7+
841
2520

z8+
23617
22680

z9

The Laurent expansion follows on dividing g by z2 and expanding the result.

> LS:=expand(g/zˆ2);

LS :=
1
z2 +

1
z

− 2 − 7 z

3
+

11 z2

6
+

79 z3

30
− z4 − 1217 z5

630
+

841 z6

2520
+

23617 z7

22680
The Laurent expansion LS converges for all values of z �= 0. By inspecting
the series and recalling that the residue is the coefficient of the 1/z term, the
residue of f at z = 0 is 1. It should be noted that the Laurent series can also
be obtained by loading the numapprox library package, applying the laurent
command to G/z2 as in check1 , and removing the order of term as in LSb.

> with(numapprox):

> check1:=laurent(G/zˆ2,z=0,N);

> LSb:=convert(check1,polynom);

LSb :=
1
z2 +

1
z

− 2 − 7 z

3
+

11 z2

6
+

79 z3

30
− z4 − 1217 z5

630
+

841 z6

2520
+

23617 z7

22680
The residue of f(z) at z = 0 can also be obtained by differentiating z2 f or G
once (since the singularity is a pole of order 2) with respect to z, evaluating the
result at z=0, and dividing by 1!

> r1:=eval(diff(G,z),z=0)/1!;

r1 := 1
An even simpler method of determining the residue of f = G/z2 at z = 0 is to
use the residue command.”

> r1b:=residue(G/zˆ2,z=0);

r1b := 1

5.4.2 Converge or Diverge?

I shall be telling this with a sigh, Somewhere ages and ages hence:
Two roads diverged in a wood, and I –
I took the one less traveled by, And that has made all the difference.
Robert Frost, American poet, The Road Not Taken (1874-1963)

Jennifer has given a second quiz question on the Laurent expansion.
Find the Laurent series for f = z2/((z − 1)2(z + 3) about z =1. Determine

the residue and the region of convergence. Confirm the latter by creating plots
for different z values which show convergence and divergence.

Again, Ms. Curious will present her recipe for solving this problem.
“The numapprox package is loaded so that the laurent command can be

used. The given function f is then entered. It has a second order pole at z=1
and a first order pole at z=−3.

> restart: with(plots): with(numapprox):

> f:=zˆ2/((z-1)ˆ2*(z+3));

f :=
z2

(z − 1)2 (z + 3)
An arrow operator L is formed for calculating the Laurent expansion of f
about z=1 to terms of order (z − 1)N . For example, for N =5, the Laurent
series has the explicit form given in LS .

> L:=N->laurent(f,z=1,N): LS:=L(5);

LS :=
1
4

(z − 1)−2 +
7
16

(z − 1)−1 +
9
64

− 9
256

(z − 1) +
9

1024
(z − 1)2

− 9
4096

(z − 1)3 +
9

16384
(z − 1)4 + O((z − 1)5)

By examining the (z − 1)−1 term, I can see that the residue is 7/16. This may
be confirmed by applying the residue command to f at z=1.

> r:=residue(f,z=1);

r :=
7
16

The region of convergence for the Laurent series is 0 < |z − 1| < 4. To test this
condition for some specified value Z of z, I first create a functional operator
gr1 which plots the Laurent expansion evaluated at z=Z over the range N =1
to 10, connecting consecutive plotting points with a thick blue straight line.

> gr1:=Z->plot([seq([N,eval(convert(L(N),polynom),z=Z)],

N=1..10)],color=blue,thickness=2):
A second operator gr2 evaluates f at z = Z and plots a horizontal line at the
calculated value over the same range of N .

> gr2:=Z->plot(eval(f,z=Z),N=1..10,thickness=2):

A third operator P superimposes gr1(Z) and gr2(Z) for a specified Z value.
> P:=Z->display({gr1(Z),gr2(Z)},labels=["N","f"]):

According to the convergence criterion given above, the Laurent expansion
should converge for z = Z = 3 and diverge for z = 7. Entering P(3) yields
the plot on the left of Figure 5.6, while P(7) generates the plot on the right.

> P(3); P(7);

0.36

0.38

0.4

f

0.42

2 4 6 8 10N

–2

0

f

2

2 4 6 8 10

N

Figure 5.6: Left plot: Convergence for z=Z =3; Right: Divergence for z=7.

The Laurent expansion clearly converges to the exact f value for z = 3 and
diverges for z=7.”

5.5 Conformal Mapping

The relation ω = f(z) describes a mapping of points in the complex z-plane
into corresponding points in the ω-plane. If f(z) is a single-valued function of
z = x + i y, each point in the z-plane maps into a single point in the ω-plane.
If f(z) is a multi-valued function with n branches, each point in the z-plane
maps into n points in the ω-plane. For example, ω=z1/2 has two branches with
each point in the z-plane producing two points in the ω-plane. The mapping
from the z-plane can be made one to one by introducing a branch cut in the
z-plane and restricting our attention to a single branch of the function, usually
referred to as the principal branch. For ω = f(z) = z1/2, the principal branch
is traditionally obtained by introducing a branch cut along the negative x-axis
and restricting the polar angle θ to the range −π < θ < π. The principal branch
of z1/2 maps only into the u > 0 half of the ω-plane, i.e., the polar angle in this
plane is restricted to the range −π/2 < θ < π/2.

If f(z) is analytic (and df/dz �= 0) at a point z0 in the z-plane, the angle be-
tween two curves intersecting at z0 is not changed on transforming to the corre-
sponding point ω0 in the ω-plane, even though the shapes of the two curves will
generally change. Transformations which have this angle-preserving property
are called conformal transformations or conformal mapping. The orthogonality
of the equipotential and field lines in potential problems is preserved under a

conformal mapping. It is not surprising therefore that a solution of Laplace’s
equation in one plane remains a solution in the other.

Conformal mapping may be used for solving 2-dimensional potential prob-
lems. Suppose that we are given a potential problem with a somewhat compli-
cated boundary condition. We then try to transform the problem to a new plane
where the boundary configuration is simpler. On solving this simpler situation,
we can transform the results back to the original plane, thus determining the
field and potential configuration for the original problem.

Many important conformal mappings, such as in the following recipe, can
be discovered by experimenting with different mathematical forms.

5.5.1 Field Around a Semi-infinite Plate

If all the ways I have been along were marked on a map and joined
up with a line, it might represent a minotaur.
Pablo Picasso, Spanish artist, (1881–1973)

In this recipe, I will use the mapping ω =
√

z to determine the equipoten-
tials and electric field (represented by field lines and by vectors) in the vicinity
of a thin semi-infinite grounded (held at zero potential) conducting plate. The
problem can be treated as 2-dimensional with the cross-section of the plate as
schematically shown in Figure 5.7.

semi-infinite plate

 infinity  -->0<--   -infinity

Figure 5.7: Cross-section of the thin semi-infinite plate.

The VectorCalculus package is loaded, because the Gradient command will
be used to calculate the electric field vector �E =−∇u, with u the potential.

> restart: with(plots): with(VectorCalculus):

The given mapping function is entered.
> w:=sqrt(z);

w :=
√

z

Before solving the problem, it is instructive to see the effect of the trans-
formation w = u + i v =

√
z =

√
x + i y on the grid lines x = c1, y = c2, where

c1 and c2 are real constants. Forming the inverse transformation z = w2 and
separating into real and imaginary parts, we have x = u2 − v2 and y = 2u v. A
grid line x= c1 is mapped into that portion of the hyperbola u2 − v2 = c1 for
which u > 0, while y=c2 is mapped into a branch of the rectangular hyperbola
2 u v=c2, the branch depending on whether c2 is positive or negative.

A graphical confirmation of the transformation of grid lines from the z-
plane to the ω-plane can be generated by applying the following conformal
command to the grid lines in a particular region of the z-plane , e.g., the region
−2 ≤ x ≤ 2, −2 ≤ y ≤ 2.

> conformal(w,z=-2-2*I..2+2*I,grid=[14,14],numxy=[100,100],

color=[red,blue],scaling=constrained,labels=["u","v"]);
The grid option is used to specify the number of grid lines in both the x
and y directions, the default being 11 × 11. The option numxy=[m,n], with
m = 100 and n = 100 here, is employed to specify the number of points to be
plotted in each grid line, with m points in the x direction and n points in the
y direction. The default is 15 points in each direction. The transformed curves
corresponding to constant x will be colored red, while those corresponding to
constant y are colored blue. The scaling is constrained and labels u, v, added.

The rectangular region −2 ≤ x ≤ 2, −2 ≤ y ≤ 2 in the z-plane maps into
the curved grid region in the ω-plane shown in Figure 5.8.

–1.5

0

1.5

v

1.5 u

Figure 5.8: Rectangular grid in z-plane maps into curved grid in ω-plane.

As expected, the curves y = c2 for c2 > 0 map into the rectangular hyperboli
in the first quadrant (u > 0, v > 0) of the w-plane, while those for c2 < 0
map into rectangular hyperboli in the fourth quadrant (u > 0, v < 0). The
x = c1 lines map into the hyperbolic curves in the figure which intersect the
rectangular hyperboli. The 90 ◦ angle between grid lines is preserved by the
conformal transformation.

Now, let’s tackle the problem of the semi-infinite grounded conducting plate.
Setting z=x + i y, ω takes the following form,

> z:=x+I*y: w:=w;

w :=
√

x + y I

which can be split into real and imaginary parts by applying the complex eval-
uation (evalc) command.

> w2a:=evalc(w) assuming y>0; w2b:=evalc(w) assuming y<0;

w2a :=

√
2
√

x2 + y2 + 2 x

2
+

1
2

I
√

2
√

x2 + y2 − 2 x

w2b :=

√
2
√

x2 + y2 + 2 x

2
− 1

2
I
√

2
√

x2 + y2 − 2 x

w2a shows the form of ω for y > 0, while w2b applies when y < 0. The potential
u relevant to our problem is obtained by removing the imaginary part from w2a.

> u:=remove(has,w2a,I);

u :=

√
2
√

x2 + y2 + 2 x

2
The electric field lines for y > 0 and y < 0 are obtained in v2a and v2b by
selecting the imaginary parts of w2a and w2b, respectively, and dividing by i.

> v2a:=select(has,w2a,I)/I: v2b:=select(has,w2b,I)/I:

The electric field vector �E =−∇u is calculated in Cartesian coordinates.
> E:=-Gradient(u,’cartesian’[x,y]);

E := −

2 x√
x2 + y2

+ 2

4
√

2
√

x2 + y2 + 2 x
ex − y

2
√

2
√

x2 + y2 + 2 x
√

x2 + y2
ey

Now, let’s plot the results. A thick blue line is plotted between (−2, 0) and
(0, 0) to represent a portion of the semi-infinite plate.

> gr1:=plot([[-2,0],[0,0]],style=line,color=blue,thickness=4):

The fieldplot command is used to plot the electric field vector �E, the vectors
being represented by thick red arrows.

> gr2:=fieldplot([E[1],E[2]],x=-2..2,y=-2..2,arrows=THICK,

grid=[10,10],color=red):
A functional operator G is formed to produce a contour plot of any input function
V , the color C of the curves to be specified. The range is taken to be x = −2
to 2 and y=−2 to 2. The contours are chosen to be V =0, 0.2, 0.4, ..., 2.

> G:=(V,C)->contourplot(V,x=-2..2,y=-2..2,contours=

[seq(0.2*n,n=0..10)],grid=[60,60],color=C):
Using G, the equipotentials (u =const.) and electric field lines are plotted
together, along with the plate, using the display command.

> display({gr1,G(u,green),G(v2a,red),G(v2b,red)});
On the computer screen, the equipotentials are colored green and the electric
field lines red. The corresponding black and white version is shown on the left
of Figure 5.9.

The semi-infinite plate is represented by the thick horizontal line between
x=−2 and 0. The equipotentials are the family of parabolas opening to the left,
the closest one to the plate being for φ=0.2, the next furthest one for φ=0.4,

and so on. The electric field lines are the family of parabolas opening to the
right. Naturally, they intersect the equipotential at right angles. This includes
the plate which is the zero equipotential. The electric field lines terminate on
the surface charge on the plate.

Figure 5.9: Equipotentials and (a) field lines (left plot), (b) field vectors (right).

The field lines convey the sense of the electric field but do not indicate how the
field strength varies with distance from the plate. To this end, the following
display command is used to superimpose the equipotentials and the electric
field ( �E) vectors.

> display({gr1,gr2,G(u,green)});
It can be seen from the resulting picture shown on the right of Figure 5.9 that
the electric field is strongest near the edge of the plate, i.e., near the tip of the
thick line in the figure.

5.5.2 A Clever Transformation

We are obliged to regard many of our original minds as crazy at least
until we have become as clever as they are.
G. C. Lichtenberg, German physicist, philosopher, (1742–99)

An infinitely long conducting cylinder of unit radius, with its axis horizon-
tal, has its upper-half held at the potential Φ = +V and its lower-half held
at Φ = −V , the two halves being separated by an infinitesimally thin layer of
insulation. What are the equipotentials and electric field inside the cylinder?

The fact that this problem is 2-dimensional in nature, and has a certain
symmetry to it, allows us to solve it rather easily by using a conformal trans-
formation approach. Loading some needed library packages,

> restart: with(plots): with(VectorCalculus):

the complex variable z = x + i y is entered. As will be seen, the mapping
w = ln((1 + z)/(1 − z)), which is inputted, will transform the original circular
boundary value problem into a planar boundary value problem which can be
solved by inspection.

> z:=x+I*y: w:=ln((1+z)/(1-z));

w := ln(
x + y I + 1

−x − y I + 1
)

The complex function w = u + i v is split into real and imaginary parts with
the complex evaluation command, and u and v separately extracted with the
remove and select commands, respectively.

> w:=simplify(evalc(w));

w :=
1
2

ln(
x2+2x+1+y2

x2−2 x+1+y2 )+arctan(
2 y

x2−2 x+1+y2 , − x2−1+y2

x2−2 x+1+y2 ) I

> u:=remove(has,w,I); v:=select(has,w,I)/I;

u :=
1
2

ln(
x2 + 2 x + 1 + y2

x2 − 2 x + 1 + y2 )

v := arctan(
2 y

x2 − 2 x + 1 + y2 , − x2 − 1 + y2

x2 − 2 x + 1 + y2 )

Recognizing that the first argument of the arctangent in v is the numerator
and the second argument the denominator, v is expressed in a more familiar
mathematical form by using the operand (op) command.

> v:=arctan(op(1,v)/op(2,v));

v := −arctan(
2 y

x2 − 1 + y2 )

Since v satisfies Laplace’s equation, it can be taken as the potential. To see
how the potential on the circular boundary transforms under the action of w,
let’s make the algebraic substitution y2 =1 − x2 in w.

> w2:=algsubs(yˆ2=1-xˆ2,w);

w2 :=
1
2

ln(
2 x + 2

−2 x + 2
) + arctan(

2 y

−2 x + 2
, 0) I

The result w2 is simplified in w3a, assuming that y > 0 and x < 1, and in w3b,
assuming that y < 0 and x < 1.

> w3a:=simplify(w2) assuming y>0,x<1;

w3a := −1
2

ln(−x + 1) +
1
2

ln(x + 1) +
1
2

I π

> w3b:=simplify(w2) assuming y<0,x<1;

w3b := −1
2

ln(−x + 1) +
1
2

ln(x + 1) − 1
2

I π

The select command is used to extract the imaginary parts of w3a and w3b
which are then divided by i in v2a and v2b, respectively.

> v2a:=select(has,w3a,I)/I; v2b:=select(has,w3b,I)/I;

v2a :=
π

2
v2b := −π

2
The upper (lower) surface of the split cylinder, which is at the potential +V
(−V ), maps into the infinite plane v2a = +π/2 (v2b = −π/2). The region
inside the split cylinder maps into the region between these infinite planes.
The potential problem in the w plane is easily solved. By inspection, the
equipotentials must be parallel planes lying between v=+π/2 and −π/2, v=0
(the u-axis) corresponding to the zero potential. Analytically, the potential
must be given by Φ = (2 v/π)V , since v = π/2 generates the equipotential
Φ=V , and so on. A picture of the equipotentials in the w plane are generated
for V =1 with the contourplot command.

> contourplot(2*Y/Pi,X=-2..2,Y=v2b..v2a);

–1

1
Y

–2 2
X

Figure 5.10: Planar equipotentials in the w-plane.

Entering Φ = 2V v/π generates the potential in terms of x and y, i.e., in the
z-plane. It can be converted to polar form if so desired, as in Φ2.

> Phi:=2*V*v/Pi;

Φ := −
2 V arctan(

2 y

x2 − 1 + y2 )

π

> Phi2:=simplify(subs({x=r*cos(theta),y=r*sin(theta)},Phi));

Φ2 := −
2 V arctan(

2 r sin(θ)
−1 + r2 )

π

The electric field expression follows on calculating �E =−∇Φ.

> E:=-Gradient(Phi,’cartesian’[x,y]);

E := − 8Vyx

(x2−1+y2)2(1+
4 y2

(x2−1+y2)2
)π

ex +
2V(

2
x2−1+y2 − 4 y2

(x2−1+y2)2
)

(1+
4 y2

(x2−1+y2)2
)π

ey

A normalized piecewise potential function is formed, made up of Φ/V inside
the unit circle and zero outside.

> pot:=piecewise(xˆ2+yˆ2<1,Phi/V,xˆ2+yˆ2>1,0);

The x and y components of the normalized electric field are also put into a
similar piecewise form in E1 and E2 .

> E1:=piecewise(xˆ2+yˆ2<1,E[1]/V,xˆ2+yˆ2>1,0):

> E2:=piecewise(xˆ2+yˆ2<1,E[2]/V,xˆ2+yˆ2>1,0):

The normalized equipotentials are plotted with the contourplot command.
> cp:=contourplot(pot,x=-1..1,y=-1..1,grid=[80,80],

contours=15):
The normalized electric field vectors are plotted using the fieldplot command.

> fp:=fieldplot([E1,E2],x=-1..1,y=-1..1,grid=[10,10],

arrows=THICK,color=red):
The two graphs, cp and fp, are superimposed to produce Figure 5.11.

> display({cp,fp},scaling=constrained);

–1

1
y

–1 1
x

Figure 5.11: Equipotentials and electric field vectors for the split cylinder.

Although this split cylinder problem was easily solved using a conformal trans-
formation, it should be noted that this approach is limited in its usefulness,
being applicable to two-dimensional potential problems of relatively simple ge-
ometry or symmetry.

5.5.3 Schwarz–Christoffel Transformation

Physical concepts are free creations of the human mind, and are not,
however it may seem, uniquely determined by the external world.
Albert Einstein, German-American physicist, Evolution of Physics, 1938

To this point, the conformal transformations have been drawn out of “thin air”.
A systematic approach for obtaining many specific transformations is to apply
the general Schwarz–Christoffel transformation which can be used to transform
the inside1 of a closed polygon in the z = x + i y plane into the upper half of
the w=u + i v plane. Referring to Figure 5.12, the vertices of the polygon are

a0

a4

a3

a2

a1

z Plane

Inside

α0

α4

α3
α2

α1

y
y

x

w Plane

InsideInside

b4b3b2b1b0

v

u

Figure 5.12: Schwarz–Christoffel transformation.

labeled a0, a1,..., the corresponding exterior angles α0, α1,..., and the trans-
forms of the vertices to the w plane b0, b1,.... Traversing the exterior of the
polygon in a counterclockwise sense, so that the interior of the polygon is to the
left, angles which correspond to turning further to the left are taken as positive,
while turns to the right are regarded as negative.

The Schwarz–Christoffel (S-C) transformation which maps the interior of
the polygon in the z plane onto the upper-half of the w plane and the boundary
of the polygon onto the real (u) axis is given by

z = A

∫
(w − b0)−α0/π(w − b1)−α1/π · · · (w − bn)−αn/π dw + B, (5.10)

where A and B are arbitrary complex constants. Note the following facts:

• Any three of the points b0, b1,...,bn may be chosen at will. We can place
b0 at infinity which removes the first factor in (5.10).

• A and B are adjusted to fix the polygon size, orientation, and position.

• Infinite open polygons are regarded as limiting cases of closed polygons.
1The method can be extended to map the outside of the polygon, see, e.g., [MF53].

As a simple example, let’s determine the S-C transformation which maps
the interior of the semi-infinite rectangular strip shown on the left of Figure 5.13
into the upper-half of the w plane. Then, this transformation will be used to
plot the equipotentials and lines of force inside a straight slit, with the same
geometry, cut in an infinite conducting sheet.

a0 ---> infinityStrip

a2 = 0

a1 = ih

α2 = π/2

α1 = π/2

y

x
0

y

1

0 x 1

Figure 5.13: Geometry (left) and equipotentials and field lines (right) for strip.

The infinite strip can be regarded as the limit of a triangle with vertices a0 → ∞,
a1 = i h, and a2 = 0. The exterior angles then are α0 = π, α1 = π/2, and
α2 = π/2. Choosing b0 = ∞, b1 = −1, and b2 = 1, the S-C transformation is
given by z=A

∫
(w + 1)−1/2(w − 1)−1/2 dw + B=

∫
(1/

√
w2 − 1) dw + B.

After loading the plots package, needed for the contour plot,
> restart: with(plots):

the integral in the S-C transformation is explicitly carried out.
> Z:=A*int(1/sqrt(wˆ2-1ˆ2),w)+B;

Z := A ln(w +
√

w2 − 1) + B

Now, when z = a2 = 0, then w = b2 = 1. Similarly, when z = a1 = i h, then
w=b1=−1. These two boundary conditions are entered,

> bc1:=eval(Z,w=1)=0; bc2:=eval(Z,w=-1)=I*h;

bc1 := B = 0 bc2 := A π I + B = h I

and solved for the constants A and B.
> sol:=solve({bc1,bc2},{A,B});

sol := {A =
h

π
, B = 0}

On assigning the solution, the S-C transformation is given by Z.
> assign(sol): Z:=Z;

Z :=
h ln(w +

√
w2 − 1)

π

w(z) may be obtained by setting Z =z, solving for ω, and simplifying.
> w:=simplify(solve(Z=z,w),symbolic);

w :=
1
2

(1 + e(
2 z π

h )) e(− z π
h )

Converting w to trig form and applying the combine command with the trig
option, yields the final simplified form of w.

> w:=combine(convert(w,trig),trig);

w := cosh(
z π

h
)

To determine the equipotentials and electric field lines in the strip, let’s enter
z=x + i y and set h=1 for plotting purposes. Then, w=u + i v is expressed in
terms of real and imaginary parts

> z:=x+I*y: h:=1: w:=evalc(w);

w := cosh(π x) cos(π y) + sinh(π x) sin(π y) I

and the forms of u and v extracted,
> u:=remove(has,w,I); v:=select(has,w,I)/I;

u := cosh(π x) cos(π y) v := sinh(π x) sin(π y)
In the w plane, the equipotentials v = const. are clearly parallel to the real
axis and the field lines u = const. perpendicular to this axis. In the z plane,
the equipotential and field lines can be obtained by using the contourplot
command, and selecting some fixed values for the contours. A function operator
cp is created to perform this task, where V will be taken to be u and v and the
color C chosen to be blue and red, respectively.

> cp:=(V,C)->contourplot(V,x=0..1,y=0..1,

[contours=seq(n,n=-4..4),-1/4,-1/2,1/4,-1/2],
grid=[100,100],color=C,thickness=2):

Using the operator cp in the following display command produces the picture
shown on the right of Figure 5.13. The conducting sheet is an equipotential so
the electric field lines in the strip intersect the edges of the sheet perpendicularly
as expected, as well as all other equipotential curves.

> display({cp(u,blue),cp(v,red)},axes=frame);

5.6 Supplementary Recipes
05-S01: Roots
Determine all the roots of z1/9, where z=(1+

√
3 i). Plot the roots and z in the

same figure, superimposing them on circles centered on the origin of radii |z|
and |z|1/9, respectively. What is the value in radians of the principal argument?

05-S02: Fluid Flow Around a Cylinder
Consider the complex function w= u + i v=V0 (z + a2/z), with z=x + i y.
(a) Determine u and v and demonstrate that they satisfy both Cauchy–

Riemann conditions and Laplace’s equation.

(b) Graphically show that v may be used to represent the streamlines outside
and perpendicular to an infinitely long rigid cylinder of radius a. Take
V0 =a=1.

(c) By calculating �V = −∇u, show that the fluid velocity components are
given by Vx = V0(−r2 + a2 cos(2 θ))/r2 and Vy = V0 a2 sin(2 θ)/r2, where
r is the radial distance from the center of the cylinder and θ is the angle
with respect to the x-axis.

(d) Calculate the fluid speed V and show that V = 0 at r = a and θ = 0, π.
These two points are called stagnation points.

05-S03: Constructing f(z)
If 5x4 y − 10 x2 y3 + y5 + 2 a2 x y/((x2 − y2)2 + 4x2 y2) is the imaginary part of
an analytic function f(z=x + i y), determine f(z).

05-S04: Analytic or Non-analytic?
Consider the function f(z)=z2 ez cos(z∗). Determine whether f is analytic or
non-analytic by (a) applying the Cauchy–Riemann conditions, (b) performing
the contour integral of f(z) along the two paths y1 = x and y2 = x2 between
z=0 and z=1 + i.

05-S05: A Contour Integral

Evaluate the integral J =
∫ ∞

−∞
((x2 − x + 2)/(x4 + 10x2 + 9)) dx using contour

integration. Confirm your answer by evaluating J with the int command.

05-S06: A Higher-order Pole

Evaluate the integral J =
∫ ∞

−∞
(x2/(x2 + a2)3) dx, with a > 0, using contour

integration. Confirm your answer by evaluating J with the int command.

05-S07: Another Angular Integral

Evaluate the integral J =
∫ π

0
(cos2(3 θ)/(5 − 4 cos(θ)) dθ using contour integra-

tion. Confirm your answer by evaluating J with the int command.

05-S08: A Removable Singularity

Evaluate the integral J =
∫ ∞

0
(sin x/x) dx using contour integration and con-

firm your answer by evaluating J with the int command. The corresponding
complex integrand sin z/z has a singularity at z =0. Because the integrand is
finite as z → 0, the singularity is called a removable singularity.

05-S09: Another Contour Integral

Evaluate the integral J =
∫ ∞

0
((x4 + 3 x2 − 2)/(x6 + 1)) dx using contour inte-

gration and check the value by using the int command.

05-S10: Fluid Flow & Electric Field Around a Plate
Consider the function w= u + i v=

√
z2 − 1, with z=x + i y.

(a) By creating a suitable figure in the x-y plane, show that the constant u
curves represent the streamlines for fluid flow around an infinitely long
plate of finite width, lying between x=−1 and +1, inserted perpendicular
to a previously uniform fluid flow in the y direction. Include the plate
and the equipotentials in your figure.

(b) Taking u to represent the equipotentials, create a plot showing the electric
field vectors and equipotentials around a thin infinite conducting plate of
finite width, lying between x=−1 and +1.

05-S11: Another Branch Cut
By taking a branch cut along the negative real axis from the branch point

x=0 to x=−∞, evaluate the integral J =
∫ ∞

0
(x1/3/(1 + x4)) dx using contour

integration. Check your answer with the int command.

05-S12: Laurent Expansion
Determine the singularities of the function f =cos(π z) e−z3

/(z2 (z − 1)5). De-
termine the Laurent series about each singularity, keeping eight terms in each
expansion. Calculate the residue of f at each singular point.

05-S13: Capacitor Edge Effects
In elementary treatments of the parallel-plate capacitor, it is assumed that the
two plates are infinitely large so that “edge effects” can be ignored. Using the
conformal command, show that the conformal transformation w=z + ez when
applied to the region −7 < x < 2, −π < y < π produces the equipotentials
and electric field lines near one edge of a finite parallel-plate capacitor. Suggest
another physical problem to which this transformation applies.



Chapter 6

Integral Transforms

Applications of the Fourier, Laplace, and Hankel transforms and their inverses
are illustrated in this chapter. The integral transform (inttrans) library pack-
age plays the key role in enabling us to perform these transforms. In some
cases, assumptions must be provided about the nature of the parameters (e.g.,
whether they are positive) in order for the transforms to be explicitly done.

6.1 Fourier Transforms

Using the Euler identity eiθ = cos θ + i sin θ, where i=
√−1, the Fourier series

representation (2.8) of the function f(x), defined over the interval −L ≤ x ≤ L,
may be expressed in the complex Fourier series form

f(x) =
∞∑

n=−∞
cn einπx/L, where cn =

1
2L

∫ L

−L

f(x) e−inπx/L dx. (6.1)

Setting kn =nπ/L and ∆kn ≡ kn+1 − kn =π/L, Eq. (6.1) may be written as

f(x) =
1
2π

∞∑
kn=−∞

∆kn

[∫ L

−L

f(y) e−ikny dy

]
eiknx. (6.2)

In the limit that L → ∞, i.e., the interval becomes infinite, Eq. (6.2) yields

f(x) =
1
2π

∫ ∞

−∞

∫ ∞

−∞
f(y) eik (x−y) dy dk, (6.3)

which is known as Fourier’s integral theorem. From (6.3), it follows that if

F (k) =
1√
2π

∫ ∞

−∞
f(x) e−ikx dx, then f(x) =

1√
2π

∫ ∞

−∞
F (k) eikx dk. (6.4)

The function F (k) is called the Fourier transform of f(x), while f(x) is the
inverse Fourier transform of F (k). It should be noted that the factor 1/(2π)
in (6.3) has been split symmetrically here. This fairly standard convention
differs from that used in Maple’s fourier and invfourier commands, where
the factor is split asymmetrically, with 1 in the Fourier transform and 1/(2π)
in the inverse. Since, in solving most problems of physical interest, usually

both the Fourier transform and its inverse are performed, this numerical factor
splitting issue normally need not concern us.

If f(x) is an even function, Equation (6.4) yields

Fc(k) =

√
2
π

∫ ∞

0
f(x) cos(kx) dx, f(x) =

√
2
π

∫ ∞

0
Fc(k) cos(kx) dk. (6.5)

Fc(k) and f(x) are the Fourier cosine transforms of each other. On the other
hand, if f(x) is an odd function one obtains the Fourier sine transforms,

Fs(k) =

√
2
π

∫ ∞

0
f(x) sin(kx) dx, f(x) =

√
2
π

∫ ∞

0
Fs(k) sin(kx) dk. (6.6)

The Fourier sine and cosine transforms are useful for solving problems involving
semi-infinite domains, while the “full” Fourier transforms (6.4) can be applied
to situations involving infinite domains.

Some of the more important properties1 of Fourier transforms are as follows:

• If the Fourier transform of f(x) (denoted by F [f(x)]) is F (k), then

F
[
dnf(x)

dxn

]
= (ik)n F (k),

provided that f(x) and its first (n − 1) derivatives vanish at x=±∞.

• The function

C(x) =
1√
2π

∫ ∞

−∞
f(x − y) g(y) dy

is called the convolution of the functions f and g (often denoted f � g)
over the interval −∞ to ∞. If F (k) and G(k) are the Fourier transforms
of f(x) and g(x), respectively, the convolution theorem is

C(x) =
1√
2π

∫ ∞

−∞
F (k) G(k) eikx dk.

Alternately, the theorem may be stated as follows: F(f � g)=F(f) F(g).

• If one sets x=0 in the convolution theorem, the Parseval relation∫ ∞

−∞
f(−y) g(y) dy =

∫ ∞

−∞
F (k) G(k) dk

results. Another Parseval relation is∫ ∞

−∞
f(y) g �(y) dy =

∫ ∞

−∞
F (k) G �(k) dk.

If g = f , this last result yields Parseval’s theorem,∫ ∞

−∞
|f(y)|2 dy =

∫ ∞

−∞
|F (k)|2 dk.

1Similar properties hold for the sine and cosine transforms.

6.1.1 Some Fourier Transform Shapes

Spoon feeding in the long run teaches us nothing
but the shape of the spoon.
E. M. Forster, British novelist, (1879–1970)

In this recipe, the Fourier transforms of some common f(x) are examined.
Calculate the Fourier transform F (k) of the following functions, using the

symmetric factor convention:
(a) f1 =A e−a2x2

; (b) f2 =A e−a |x|; (c) f3 =A for |x| < a, 0 for |x| > a.
Plot the f(x) and their transforms for a=A=1. Confirm Parseval’s theorem.

The inttrans library package contains the Fourier transform command,
fourier, so is loaded. To accomplish the transforms, the assumption that both
a and A are positive is entered.

> restart: with(inttrans): assume(a>0,A>0):

A functional operator F is formed to take the Fourier transform of an arbitrary
function f and simplify the result. The result is divided by

√
2π to agree with

the symmetric factor convention.
> F:=f->simplify(fourier(f,x,k)/sqrt(2*Pi)):

The Gaussian function f1 is entered, and its Fourier transform F1 calculated
using the operator F. The transform is a Gaussian profile centered on k=0.

> f1:=A*exp(-aˆ2*xˆ2); F1:=F(f1);

f1 := A e(−a2 x2) F1 :=
1
2

A e(− k2

4 a2 ) √
2

a
The symmetric exponentially decaying function f2 is entered and its transform
calculated. The resulting profile F2 is a Lorentzian line shape centered on k=0.

> f2:=A*exp(-a*abs(x)); F2:=F(f2);

f2 := A e(−a |x|) F2 :=
A a

√
2

(a2 + k2)
√

π
Making use of the Heaviside command, the rectangular profile f3 is entered,
and its transform F3 determined.

> f3:=A*Heaviside(x+a)-A*Heaviside(x-a); F3:=F(f3);

f3 := A Heaviside(x + a) − A Heaviside(x − a)

F3 :=
A sin(a k)

√
2

k
√

π
To plot each function and its transform, an arrow operator G is introduced to
evaluate an arbitrary function FF at a=1, A=1.

> G:=FF->eval(FF,{a=1,A=1}):
Making use of G, the three given functions f1 , f2 , and f3 are plotted together,
the curves being given different colors and line styles. The resulting picture is
shown on the left of Figure 6.1.

> plot([G(f1),G(f2),G(f3)],x=-4..4,color=[red,blue,green],

linestyle=[1,2,3],thickness=2,labels=["x","f"]);

0

f

1

–4 x 4

F

0.8

–8 x 8

Figure 6.1: Left: the three f(x). Right: Corresponding Fourier transforms.

The three Fourier transforms F1 , F2 , and F3 are plotted with matching colors
and line styles and shown on the right of the figure.

> plot([G(F1),G(F2),G(F3)],k=-10..10,color=[red,blue,green],

linestyle=[1,2,3],thickness=2,labels=["x","F"]);
The oscillatory curve is the transform of the rectangle f3 , the shorter smooth
curve the transform of the Gaussian f1 , and the taller smooth curve is the
transform F2 . The correspondence is more evident on the computer screen.

Using the complex conjugate (conjugate) command, an operator L is formed
to calculate the left-hand side of Parseval’s theorem for a given function f .

> L:=f->int(f*conjugate(f),x=-infinity..infinity);

L := f →
∫ ∞

−∞
f (f) dx

Similarly, an operator R is created to calculate the right-hand side of Parseval’s
theorem for a specified Fourier transform FT .

> R:=FT->int(FT*conjugate(FT),k=-infinity..infinity);

R := FT →
∫ ∞

−∞
FT (FT ) dk

Using L and R, Parseval’s theorem is now confirmed for the function f1 and its
transform F1 , the results L1 and R1 being identical.

> L1:=L(f1); R1:=R(F1);

L1 :=
A2 √

π
√

2
2 a

R1 :=
A2 √

π
√

2
2 a

It is left as an exercise for you to confirm that Parseval’s theorem holds for the
other two cases.

6.1.2 A Northern Weenie Roast

Wedding: the point at which a man stops toasting a woman
and begins roasting her.
Helen Rowland, American journalist, A Guide to Men,“Syncopations” (1922)

While visiting his inlaws in Northern Ontario over the Christmas holidays,
Russell went snowmobiling with his family. After a few hours, feeling slightly
cold and hungry, they stopped at one of the warming huts along the track and
soon had a roaring fire going in the fireplace. Using some long thin steel rods
which had been conveniently leaning against the outer wall of the hut, they
satisfied their hunger by roasting weenies in the fire.

Idealizing the cooking situation, let’s calculate and animate the temperature
distribution inside a thin insulated semi-infinite (0 ≤ x ≤ ∞) steel rod which
has the x=0 end held at a fixed temperature T0 and whose interior is initially
at 0 degrees Celsius. For steel, the thermal conductivity K =46 W/(m K◦), the
density ρ=8 × 103 kg/m3, and the specific heat C =500 J/(kg K◦).

The plots and integral transform packages are loaded, the former needed for
the animation, the latter for the Fourier sine (fouriersin) command.

> restart: with(plots): with(inttrans):

The one-dimensional heat diffusion equation is entered in de, T (x, t) being
the temperature at a distance x from the hot end of the rod at time t. The
parameter a =

√
K/(ρC).

> de:=diff(T(x,t),t)=aˆ2*diff(T(x,t),x,x);

de :=
∂

∂t
T(x, t) = a2 (

∂2

∂x2 T(x, t))

Since the temperature is specified at x=0, i.e., T (0, t)=T0, we will solve the
diffusion equation by first taking the Fourier sine transform of de, yielding the
time-dependent ODE shown in the output of eq .

> eq:=fouriersin(de,x,k);

eq :=
∂

∂t
fouriersin(T(x, t), x, k) =

a2 k (
√

2 T(0, t) − k fouriersin(T(x, t), x, k)
√

π)√
π

The notation is simplified by replacing fouriersin(T(x, t), x, k) with F(k, t).
> eq2:=subs(fouriersin(T(x,t),x,k)=F(k,t),eq);

eq2 :=
∂

∂t
F(k, t) =

a2 k (
√

2 T(0, t) − k F(k, t)
√

π)√
π

The boundary condition T (0, t) = T0 is substituted into eq2 .
> eq3:=subs(T(0,t)=T0,eq2);

eq3 :=
∂

∂t
F(k, t) =

a2 k (
√

2T0 − k F(k, t)
√

π)√
π

Then, the ODE eq3 is analytically solved for F(k, t).
> eq4:=dsolve(eq3,F(k,t));

eq4 := F(k, t) =
√

2T0
k

√
π

+ e(−a2 k2 t) F1(k)

Since T (x, 0)=0 for x > 0, then F (k, 0)=0. So, the arbitrary function F1(k)
is determined by evaluating the right-hand side of eq4 at time t=0, setting the
result equal to 0, and solving for F1(k).

> _F1(k):=solve(eval(rhs(eq4),t=0),_F1(k));

F1(k) := −
√

2T0
k

√
π

F1(k) is automatically substituted into eq4 , the result being labeled eq5 .
> eq5:=eq4;

eq5 := F(k, t) =
√

2T0
k

√
π

− e(−a2 k2 t)
√

2T0
k

√
π

The temperature distribution inside the rod at arbitrary time t ≥ 0 follows on
applying the following Fourier sine command to the right-hand side of eq5 , and
simplifying with the symbolic option.

> T:=simplify(fouriersin(rhs(eq5),k,x),symbolic);

T := −T0 (−1 + erf(
x

2 a
√

t
))

The temperature T is expressed in terms of the error function. As a partial check
on our calculation, note that for x > 0 and t = 0, T (x, 0) =T0 (1 − erf(∞)) =
T0 (1 − 1) = 0 as it should. The thermal conductivity (K), density (ρ), and
specific heat (C) values for steel are now entered and the diffusion coefficient
a =

√
K/(ρ C) calculated. In the MKS system, a has the units m/s1/2. For

convenience, a is converted to CGS units by multiplying by 100, since 1 m =
100 cm. The units of a are then cm/s1/2, the distance x being expressed in cm.

> K:=46: rho:=8*10ˆ3: C:=500: a:=100*sqrt(K/(rho*C));

a :=
√

46
20

The normalized temperature distribution T/T0 is given in TT ,
> TT:=T/T0;

TT := 1 − erf(
5 x

√
46

23
√

t
)

which is plotted at 60 second (1 minute) time intervals up to 5 minutes.
> plot([seq(eval(TT,t=60*i),i=1..5)],x=0..25,thickness=2,

labels=["x","T/T0"]);
The resulting time evolution of the temperature inside the rod is shown in
Figure 6.2, the curve furthest to the left corresponding to 1 minute, the curve
furthest to the right to 5 minutes. Even at 5 minutes the temperature at x=25
cm is a tiny fraction of the temperature at the hot end.

Figure 6.2: Normalized temperature distribution inside rod at different times.

Clearly, the semi-infinite approximation to the long rod is not too bad. The nor-
malized temperature distribution is now animated over the same time interval,
100 frames being used.

> animate(TT,x=0..25,t=0..300,frames=100,thickness=2,

labels=["x","T/T0"]);
You will have to execute the recipe on your computer to see the animation.

6.1.3 Turn Off the Boob Tube and Concentrate

Television’s perfect. You turn a few knobs, a few of those mechanical
adjustments at which the higher apes are so proficient, and lean back
and drain your mind of all thought. And there you are watching the
bubbles in the primeval ooze. You don’t have to concentrate. You
don’t have to react. You don’t have to remember. You don’t miss
your brain because you don’t need it.
Raymond Chandler, American novelist, (1888–1959)

Since you are reading this book, you have turned your back (at least tem-
porarily) on watching TV. Having your undivided attention, I would like you
to concentrate on the following example involving the Fourier cosine transform.

Consider a thin semi-infinite (0 ≤ x ≤ ∞) hollow tube, closed at the x=0
end, containing water. A drop of blue ink is injected at the point x = x0 > 0
at time t=0. Suppose that the drop has an initial concentration profile given
approximately by C(x, 0)=f(x)=A δ(x−x0). If the diffusion constant of ink in
water is d, determine the concentration of ink for arbitrary time t ≥ 0. At what
time T is the concentration of ink at x = 0 a maximum? Taking the nominal
values A = 1, d = 1, and x0 = 10, evaluate T and animate the concentration
profile C(x, t) and plot C(0, t) over a time interval up to 2T .

The plots and integral transform packages are loaded and the 1-dimensional
diffusion equation entered in pde for the concentration C(x, t) of ink.

> restart: with(plots): with(inttrans):

> pde:=diff(C(x,t),t)=d*diff(C(x,t),x,x);

pde :=
∂

∂t
C(x, t) = d (

∂2

∂x2 C(x, t))

Since there can be no flow of ink through the x = 0 end, the concentration
gradient ∂C/∂x must be zero there. This boundary condition implies that we
should use a Fourier cosine transform approach on the diffusion equation. The
Fourier cosine transform (fouriercos) command is applied to pde.

> eq:=fouriercos(pde,x,k);

eq :=
∂

∂t
fouriercos(C(x, t), x, k) =

−d (
√

2 D1(C)(0, t) + k2 fouriercos(C(x, t), x, k)
√

π)√
π

The result in eq is expressed in terms of the concentration gradient at x = 0,
viz., D1(C)(0, t). This term is set equal to zero in eq2 . The notation is also
simplified by replacing fouriercos(C(x, t), x, k) with F(k, t).

> eq2:=subs({fouriercos(C(x,t),x,k)=F(k,t),D[1](C)(0,t)=0},eq);

eq2 :=
∂

∂t
F(k, t) = −d k2 F(k, t)

The initial concentration profile f =A δ(x − x0) of the ink is entered,
> f:=A*Dirac(x-x0);

f := A Dirac(−x + x0 )
and its Fourier cosine transform taken and simplified assuming that x0 > 0.

> F1:=simplify(fouriercos(f,x,k)) assuming x0>0;

F1 :=
A

√
2 cos(k x0 )√

π
The ODE eq2 is analytically solved for F(k, t).

> eq3:=dsolve(eq2,F(k,t));

eq3 := F(k, t) = F1(k) e(−d k2 t)

At t=0, F(k, 0)= F1(k), so F1(k)=F1 is substituted into eq3 .
> eq4:=subs(_F1(k)=F1,eq3);

eq4 := F(k, t) =
A

√
2 cos(k x0 ) e(−d k2 t)

√
π

The general desired expression for the concentration follows on applying the
Fourier cosine transform to the right-hand side of eq4 , taking us from k-space
back into x-space.

> C:=fouriercos(rhs(eq4),k,x);

C :=
A

√
π

t d
e(− x02+x2

4 t d ) cosh(
x0 x

2 t d
)

π
The time T at which the ink concentration is an extremum at x=0 is obtained
by evaluating C at this point, differentiating with respect to t, setting the result
equal to zero, and solving for the time t. The plot of the concentration at x=0
will confirm that the time T corresponds to the maximum ink concentration at
that end of the tube.

> T:=solve(diff(eval(C,x=0),t)=0,t);

T :=
x0 2

2 d
The parameter values A=1, d=1, and x0=10 are entered and the concentration
profile C(x, t) and time T then determined.

> A:=1: d:=1: x0:=10: C:=C; T:=T;

C :=

√
π

t
e(− 100+x2

4 t ) cosh(
5 x

t
)

π

T := 50
The concentration reaches its maximum value at the x = 0 end 50 time units
after the ink drop is injected.

The concentration profile is animated over the time interval t=1 to 2T , 100
frames being taken. The animation does not start at t = 0 because the initial
profile is a Dirac delta function. The initial frame is shown in Figure 6.3. Run
the animation and see how the profile evolves with time.

> animate(C,x=0..35,t=1..2*T,frames=100,thickness=2,

numpoints=200,labels=["x","C"]);

0

0.05

0.1

C

0.2

0.25

5 10 15 x 25 30 35

Figure 6.3: Initial frame in the animation of the ink concentration.

The concentration of ink at x = 0 is now plotted in Figure 6.4 over the time
interval t=0 to 2 T . The maximum concentration occurs at 50 time units.

> plot(eval(C,x=0),t=0..2*T,thickness=2,

labels=["t","C(0,t)"]);

0

0.01

0.02

C(0,t)

0.04

20 40 t 80 100

Figure 6.4: Time evolution of the ink concentration at x=0.

6.1.4 Diffusive Heat Flow

In my own time there have been inventions of this sort,...tubes for
diffusing warmth equally through all parts of a building...
Seneca, Roman writer, philosopher, and statesman, (5BC–65AD)

In contrast to the radiant heating referred to by Seneca, involving the con-
vective flow of warm water in tubes under the floor, this recipe involves the
diffusive flow of heat in an infinitely long thin solid rod. Suppose that the
initial temperature distribution inside the rod is T (x, 0) = T0 e−b2x2

. What is
the temperature distribution T (x, t) for times t ≥ 0? Animate the suitably
normalized temperature distribution.

Since the domain is infinite, the full Fourier transform will be used. Because
the inverse will also be calculated, it is not necessary to worry about the fact
that the Maple convention for splitting the factor 1/(2π) in Fourier’s integral
theorem differs from the convention adopted in this chapter. As before, the
integral transform and plots (needed for the animation) packages are loaded.

> restart: with(plots): with(inttrans):

In order for the transforms to be explicitly calculated, the assumption that
b > 0, t > 0, and the heat diffusion coefficient a > 0, is supplied.

> assume(a>0,b>0,t>0):

The one-dimensional heat diffusion equation is entered.

> de:=diff(T(x,t),t)=aˆ2*diff(T(x,t),x,x);

de :=
∂

∂t
T(x, t) = a2 (

∂2

∂x2 T(x, t))

The Fourier transform of the PDE de is taken in eq and notationally simplified
in eq2 by substituting F(k, t) for fourier(T(x, t), x, k).

> eq:=fourier(de,x,k);

eq :=
∂

∂t
fourier(T(x, t), x, k) = −a2 k2 fourier(T(x, t), x, k)

> eq2:=subs(fourier(T(x,t),x,k)=F(k,t),eq);

eq2 :=
∂

∂t
F(k, t) = −a2 k2 F(k, t)

Using the dsolve command, the ODE eq2 is analytically solved for F(k, t).
> sol:=dsolve(eq2,F(k,t));

sol := F(k, t) = F1(k) e(−a2 k2 t)

The arbitrary function F1(k) is determined by Fourier transforming the initial
temperature profile, T0 e−b2x2

. Its functional form is automatically substituted
into sol , the result being labeled sol2 .

> _F1(k):=fourier(T0*exp(-bˆ2*xˆ2),x,k); sol2:=sol;

F1(k) := T0 e(− k2

4 b2 )
√

π

b2

sol2 := F(k, t) = T0 e(− k2

4 b2 )
√

π

b2 e(−a2 k2 t)

The temperature profile inside the rod for t > 0 follows on taking the inverse
Fourier transform (invfourier) of the right-hand side of sol2 .

> T:=invfourier(rhs(sol2),k,x);

T :=
T0 e

(− b2 x2

4 b2 a2 t+1 )
√

4 b2 a2 t + 1
The temperature is normalized (made dimensionless) by dividing T by T0 , and
normalized spatial (ζ) and time (τ) coordinates are introduced by substituting
x=ζ/b and t=τ/(b2a2).

> Tnorm:=subs({x=zeta/b,t=tau/(bˆ2*aˆ2)},T/T0);

Tnorm :=
e
(− ζ2

4 τ+1 )
√

4 τ + 1
The normalized temperature, Tnorm, is animated over the time interval τ =0
to 10, the spatial range being ζ =−10 to 10.

> animate(Tnorm,zeta=-10..10,tau=0..10,frames=100,

thickness=2,numpoints=200,labels=["zeta","Tnorm"]);
Run the animation to see the initially Gaussian temperature profile diffuse away.

6.1.5 Deja Vu

The postmodern reply to the modern consists of recognizing that the
past, since it cannot really be destroyed, because its destruction leads
to silence, must be revisited...
Umberto Eco, Italian novelist, Reflections on the Name of the Rose, (1983)

If this example seems familiar, it is. We revisit Recipe 01-1-6: Mr. Dirac’s
Famous Function and, instead of using the Green’s function method, Fourier
transform the relevant differential equation and illustrate how contour integra-
tion can be used to evaluate the inverse transform.

Consider a very long uniform horizontal beam glued to an elastic foundation
which exerts a Hooke’s law restoring force (spring constant K) on the beam.
Subject to a steady point force F exerted at x = 0, the static displacement
ψ(x)=(F/K) G(x) of the beam satisfies the following ODE,

α4 d4G

dx4 + G = δ(x), (6.7)

with α = (Y I/K)1/4, Y being Young’s modulus and I the beam’s moment of
inertia. Determine G(x) for all x.

After loading the integral transform package, the following interface com-
mand is used to set j =

√−1, to avoid confusion with the moment of inertia.

> restart: with(inttrans): interface(imaginaryunit=j):

Equation (6.7) is entered in ode and then Fourier transformed in eq .

> ode:=alphaˆ4*diff(G(x),x$4)+G(x)=Dirac(x);

ode := α4 (
d4

dx4 G(x)) + G(x) = Dirac(x)

> eq:=fourier(ode,x,k);

eq := fourier(G(x), x, k) (α4 k4 + 1) = 1
eq is solved in eq2 for the Fourier transform of G(x).

> eq2:=solve(eq,fourier(G(x),x,k));

eq2 :=
1

α4 k4 + 1
With Maple’s convention for the Fourier transform, the inverse Fourier trans-
form is given by G(x)=

∫∞
−∞ ejkx(eq2/(2π)) dk. The integrand is now entered.

> integrand:=exp(j*k*x)*eq2/(2*Pi);

integrand :=
1
2

e(k x j)

(α4 k4 + 1)π

The four complex k zeros of the integrand’s denominator are determined.

> sol:={solve(denom(integrand)=0,k)};

sol :=

⎧⎪⎪⎨
⎪⎪⎩

√
2

2
+

1
2

j
√

2

α
,

1
2

j
√

2 −
√

2
2

α
,

−
√

2
2

− 1
2

j
√

2

α
,

−1
2

j
√

2 +
√

2
2

α

⎫⎪⎪⎬
⎪⎪⎭

The integrand has four simple poles, the first two entries in sol corresponding
to poles in the upper-half of the complex k-plane, the last two entries to poles in
the lower-half k-plane. The integral in G(x) is evaluated by performing a closed
contour integration along the real k-axis from −R to +R and then around a
semi-circle of radius R (where k=R ejθ), the limit R → ∞ being taken. Noting
that on the semi-circle the exponential term in the integrand takes the form
ejRejθx = ejR cos(θ)x e−R sin(θ)x, the semi-circle must be taken in the upper-half
k-plane (where sin(θ) > 0) for x > 0 for the semi-circular contribution to vanish
as R → ∞. In this case, the two poles in the upper-half plane will contribute
to the integral.

For x < 0, on the other hand, the semi-circle must be taken in the lower-half
plane where sin(θ) < 0. The two poles in the lower-half k-plane then contribute.

A functional operator is formed for evaluating the residue of the integrand
for the ith simple pole.

> res:=i->simplify(evalc(residue(integrand,k=sol[i]))):

Using the functional operator, the sequence of four residues is explicitly calcu-
lated, the first two residues corresponding to the two poles in the upper-half
plane, the last two residues to the two poles in the lower-half plane.

> eq3:={seq(res(i),i=1..4)};
eq3 :=

{
− 1

16

e(−x
√

2
2 α ) √

2 (cos(
x

√
2

2 α
) − sin(

x
√

2
2 α

) + sin(
x

√
2

2 α
) j + cos(

x
√

2
2 α

) j)

α π
,

− 1
16

e(−x
√

2
2 α ) √

2 (−cos(
x

√
2

2 α
) + sin(

x
√

2
2 α

) + sin(
x

√
2

2 α
) j + cos(

x
√

2
2 α

) j)

α π
,

1
16

e(
x

√
2

2 α ) √
2 (cos(

x
√

2
2 α

) + sin(
x

√
2

2 α
) − sin(

x
√

2
2 α

) j + cos(
x

√
2

2 α
) j)

α π
,

1
16

e(
x

√
2

2 α ) √
2 (−cos(

x
√

2
2 α

) − sin(
x

√
2

2 α
) − sin(

x
√

2
2 α

) j + cos(
x

√
2

2 α
) j)

α π

}

The last two residues, containing ex
√

2/(2α), are selected in eq4 , while the first
two residues, containing e−x

√
2/(2α), are selected in eq5 .

> eq4:=select(has,eq3,exp(x*sqrt(2)/(2*alpha))):

> eq5:=select(has,eq3,exp(-x*sqrt(2)/(2*alpha))):

For x < 0 (to the left of the applied force at the origin), then G(x) is obtained
by summing the two residues in eq4 , multiplying by 2 j π, applying the complex
evaluation command, and simplifying. The minus sign is inserted because for
the semi-circle in the lower-half k-plane, the contour is taken clockwise, rather
than counter-clockwise. The result, assigned the name GL, is the same as
obtained previously in Recipe 01-1-6.

> GL:=-simplify(evalc(2*j*Pi*(eq4[1]+eq4[2])));

GL :=
1
4

e(
x

√
2

2 α ) √
2 (−sin(

x
√

2
2 α

) + cos(
x

√
2

2 α
))

α
The solution for x > 0 is similarly obtained by using eq5 . The contour is
counter-clockwise in this case, so a minus sign is not inserted. The result GR
is identical with that previously obtained using the Green’s function method.

> GR:=simplify(evalc(2*j*Pi*(eq5[1]+eq5[2])));

GR :=
1
4

e(−x
√

2
2 α ) √

2 (cos(
x

√
2

2 α
) + sin(

x
√

2
2 α

))

α

6.2 Laplace Transforms

The Laplace transform of a function f(t) is defined to be

L(f(t)) = F (s) =
∫ ∞

0
e−st f(t) dt, (6.8)

where s=x + i y is a complex variable. The inverse transform is given by the
Bromwich integral formula,

L−1(F ) = f(t) =
1

2πi

∫ c+i ∞

c−i ∞
est F (s) ds, t > 0 (6.9)

and f(t)=0 for t < 0. The integration in (6.9) is to be performed along a line
s=c+ i y in the complex plane. The real number c is chosen so that s=c lies to
the right of all poles, branch points, or essential singularities. With Maple, the
laplace command is used to calculate the Laplace transform, and invlaplace
for the inverse transform.

A couple of the more important properties of Laplace transforms are:

• The Laplace transform of the nth derivative of f(t) is given by

L
(

dnf

dtn

)
= sn F (s) −

n−1∑
k=0

(
dkf

dtk

)
t=0

sn−k−1.

• Defining the convolution of the functions f1(t) and f2(t) to be

C(t) =
∫ t

0
f1(t − y) f2(y) dy, then L(C(t)) = F1(s) F2(s),

where F1(s) and F2(s) are the Laplace transforms of f1(t) and f2(t).

6.2.1 Jennifer Consults Mr. Spiegel

Histories are more full of examples of the fidelity of dogs
than of friends.
Alexander Pope, English satirical poet, (1688–1744).

To illustrate the use of Laplace transforms in solving ODEs, Jennifer has cre-
ated a recipe for solving the following problem taken from Murray Spiegel’s
Advanced Mathematics. Solve the fourth order inhomogeneous ODE

y ′′′′(t) + 2 y ′′(t) + y(t) = sin t, y(0) = 1, y ′(0) = −2, y ′′(0) = 3, y ′′′(0) = 0.

After loading the integral transform package, Jennifer lets the Laplace transform
of y(t) be represented by F (s) for notational convenience.

> restart: with(inttrans): LT:=laplace(y(t),t,s)=F(s):

The ODE is now entered.
> ode:=diff(y(t),t$4)+2*diff(y(t),t$2)+y(t)=sin(t);

ode := (
d4

dt4
y(t)) + 2 (

d2

dt2
y(t)) + y(t) = sin(t)

She then takes the Laplace transform of ode and substitutes LT ,
> eq:=subs(LT,laplace(ode,t,s));

eq := s4 F (s) − (D(3))(y)(0) − s (D(2))(y)(0) − s2 D(y)(0) − s3 y(0)

+ 2 s2 F (s) − 2 D(y)(0) − 2 s y(0) + F (s) =
1

s2 + 1
the result being expressed in terms of y(0) and the first three derivatives at
t=0. Maple has simply applied the rule for taking the Laplace transform of a
derivative to the fourth and second order derivatives. The last term, 1/(s2 +1),
in eq is the Laplace transform of sin(t). If proceeding by hand, one would look
up this result in a table of Laplace transforms.

The initial conditions are now entered in ic, using the differential operator,
> ic:=(y(0)=1,D(y)(0)=-2,D(D(y))(0)=3,D(D(D(y)))(0)=0):

and eq is evaluated in eq2 with the initial conditions.
> eq2:=eval(eq,{ic});

eq2 := s4 F (s) + 4 − 5 s + 2 s2 − s3 + 2 s2 F (s) + F (s) =
1

s2 + 1
Jennifer then solves eq2 for F (s).

> eq3:=solve(eq2,F(s));

eq3 :=
−6 s2 − 3 + 6 s3 + 5 s − 2 s4 + s5

s6 + 3 s4 + 3 s2 + 1
The solution, Y , of the original ODE then follows on applying the inverse
Laplace transform to eq3 . If proceeding by hand, as in Spiegel, one would
laboriously express eq3 as a sum of partial fractions and then use a table to
look up the inverse transforms of the various terms.

> Y:=invlaplace(eq3,s,t);

Y := −1
8

sin(t) (21 + t2 − 16 t) +
1
8

cos(t) (8 + 5 t)

Exactly the same answer can be obtained more easily by using the dsolve
command with the option method=laplace, as illustrated in Y2 .

> Y2:=dsolve({ode,ic},y(t),method=laplace);

Y2 := y(t) = −1
8

sin(t) (21 + t2 − 16 t) +
1
8

cos(t) (8 + 5 t)

Finally, Jennifer completes the recipe by plotting Y over the time interval t=0
to 100, the solution being shown in Figure 6.5.

> plot(Y,t=0..100,thickness=2,labels=["t","y"],numpoints=200);

t

Figure 6.5: Time-dependent solution of the ODE.

As time increases, the oscillations grow indefinitely in amplitude, never settling
down to a steady-state solution.

6.2.2 Jennifer’s Heat Diffusion Problem

Gossip is a sort of smoke that comes from the dirty ...pipes of those
who diffuse it: it proves nothing but the bad taste of the smoker.
George Eliot, English novelist, (1819–80)

The Laplace transform approach can be used to solve the diffusion equation
with specified boundary conditions and an initial profile. Jennifer will illustrate
this by solving the following temperature diffusion problem and animating it
over the time interval t=0 to 0.05 seconds,

Tt =4 Txx, T (0, t)=T (3, t)=0, T (x, 0)=10 sin(2πx) − 6 sin(4πx) + 2 sin(6πx).

After loading the integral transform and plots packages,
> restart: with(inttrans): with(plots):

Jennifer enters the relevant PDE and the initial temperature profile T (x, 0).
> pde:=diff(T(x,t),t)=4*diff(T(x,t),x,x);

pde :=
∂

∂t
T (x, t) = 4 (

∂2

∂x2 T (x, t))

> T(x,0):=10*sin(2*Pi*x)-6*sin(4*Pi*x)+2*sin(6*Pi*x);

Then pde is Laplace transformed in eq with respect to the time variable t, the
initial condition being automatically substituted. For notational simplicity the
Laplace transform of T (x, t) is replaced with F (x), the resulting ODE for F (x)
being displayed in eq2 .

> eq:=laplace(pde,t,s);

> eq2:=subs(laplace(T(x,t),t,s)=F(x),eq);

eq2 := sF (x) − 10 sin(2π x) + 6 sin(4π x) − 2 sin(6π x) = 4 (
d2

dx2 F (x))

Since the boundary conditions are T (0, t) = T (3, t) = 0, their Laplace trans-
forms are equal to zero. So eq2 is analytically solved for F (x) subject to
F (0) = 0 and F (3) = 0.

> sol:=dsolve({eq2,F(0)=0,F(3)=0},F(x));

sol := F (x) = ((160 s π2+2 s2+2048π4) sin(6π x) − 13824((
s

16
+π2) sin(4π x)

−20
3

sin(2π x) (
s

64
+π2)) (

s

144
+π2))

/
(s3+224 s2 π2+12544 s π4+147456π6)

Finally, applying the inverse Laplace transform to the right-hand side of sol
yields the temperature profile T (x, t) for t ≥ 0, which is animated over the time
interval t=0 to 0.05.

> T(x,t):=invlaplace(rhs(sol),s,t);

T (x, t) := 10 sin(2πx) e(−16 π2 t) − 6 sin(4πx) e(−64 π2 t) + 2 sin(6πx) e(−144 π2 t)

> animate(T(x,t),x=0..3,t=0..0.05,frames=100,numpoints=500,

thickness=2,labels=["x","T"]);

Figure 6.6: Initial frame of the animation.

The initial frame of the animation is shown in Figure 6.6, with 100 frames
being taken and 500 plotting points used to produce a smooth figure. The
temperature distribution becomes effectively zero throughout the entire range
x=0 to 3 on completion of the animation.

6.2.3 Daniel Strikes Yet Again: Mr. Laplace Appears

Propaganda is a soft weapon; hold it in your hands too long,
and it will move about like a snake, and strike the other way.
Jean Anouilh, French playwright, (1910–1987)

In Example 4.1.3, we imagined a scenario where young Daniel struck a light,
elastic, initially horizontal, string of length L fixed at both ends. Our approach
to determining the subsequent motion of the string was to build up a solution
using a Fourier series approach. In the present recipe, a similar struck string
is considered but now solved for the transverse displacement U(x, t) using the
Laplace transform method. In particular, we take L = π m, wave speed c = 2
m/s, and an initial transverse velocity ∂U(x, 0)/∂t = 3 sin(2 x) − 2 sin(5 x).
After solving for U(x, t), the motion of the string is animated over the time
interval t=0 to T =5 s.

After loading the integral transform and plots packages,
> restart: with(inttrans): with(plots):

the values of c, L, and T are entered.
> c:=2: L:=Pi: T:=5:

To simplify the notation, the addtable command is used to let F (x) represent
the Laplace transform of U(x, t) with respect to t in subsequent outputs.

> addtable(laplace,U(x,t),F(x),t,s):

The wave equation for U(x, t) is entered in pde, the value of c being automati-
cally substituted.

> pde:=diff(U(x,t),x,x)=(1/cˆ2)*diff(U(x,t),t,t);

pde :=
∂2

∂x2 U (x, t) =
1
4

(
∂2

∂t2
U (x, t))

The Laplace transform of pde with respect to time t is taken in eq .
> eq:=laplace(pde,t,s);

eq :=
d2

dx2 F (x) =
1
4

s2 F (x) − 1
4

D2(U)(x, 0) − 1
4

sU (x, 0)

The output in eq is given in terms of the initial profile (U (x, 0)) and the initial
transverse velocity (D2(U)(x, 0), with D2 indicating a time derivative). These
are supplied in the initial condition, ic, which is substituted into eq in eq2 .

> ic:=U(x,0)=0,D[2](U)(x,0)=3*sin(2*x)-2*sin(5*x);

ic := U (x, 0) = 0, D2(U)(x, 0) = 3 sin(2x) − 2 sin(5x)
> eq2:=subs(ic,eq);

eq2 :=
d2

dx2 F (x) =
1
4

s2 F (x) − 3
4

sin(2x) +
1
2

sin(5x)

The ODE eq2 is analytically solved for F (x), subject to the two boundary
conditions, F (0)=0 and F (L)=0

> sol:=dsolve({eq2,F(0)=0,F(L)=0},F(x));
sol := F (x) = −2(16 cos(x)4 s2 + 256 cos(x)4 − 192 cos(x)2 − 12 cos(x)2 s2

−3 s2 cos(x) − 300 cos(x) + s2 + 16)sin(x)/(s4 + 116 s2 + 1600)
The solution U of the wave equation follows on taking the inverse Laplace
transform of the right-hand side of sol .

> U:=invlaplace(rhs(sol),s,t);

U :=
1
10

sin(x) (15 cos(x) sin(4 t) + 2 sin(10 t) (4 cos(x)2 − (4 cos(x)2 − 1)2))

U can be simplified by applying the combine command with the trig option.
> U:=combine(U,trig);

U :=
1
10

cos(5 x+10 t) − 1
10

cos(−5 x+10 t) − 3
8
cos(2 x+4 t) +

3
8
cos(−2 x+4 t)

The motion of the string is now animated over the time interval t=0 to T , with
100 frames being taken and 500 plotting points used.

> animate(U,x=0..L,t=0..T,frames=100,numpoints=500,

thickness=2,labels=["x","U"]);
You will have to run the animation to see how the string vibrates.

6.2.4 Infinite-medium Green’s Function

Private information is practically the source
of every large modern fortune.
Oscar Wilde, Anglo-Irish playwright, author, (1854–1900)

In this recipe we consider the 3-dimensional infinite-medium diffusion equa-
tion (diffusion coefficient d) with a point source of unit magnitude located at
�r = (x, y, z) = 0 and turned on at the instant t = 0. The relevant PDE for the
Green’s function G(�r, t) is of the form

∂G

∂t
− d ∇2G = δ(�r) δ(t). (6.10)

Prior to the switching on of the source G = 0 everywhere. Our goal is to
determine G(x, y, z, t) for t > 0 and animate the solution.

In addition to the usual integral transform and plots packages that have
appeared regularly in the recipes of this chapter, the VectorCalculus package is
also loaded because the Laplacian command will be used to enter ∇2 in (6.10).

> restart: with(inttrans): with(plots): with(VectorCalculus):

Eq. (6.10) is entered in Cartesian coordinates, noting that δ(�r)=δ(x) δ(y) δ(z).

> pde:=diff(G(x,y,z,t),t)-d*Laplacian(G(x,y,z,t),’cartesian’

[x,y,z])=Dirac(x)*Dirac(y)*Dirac(z)*Dirac(t);

pde := (
∂

∂t
G(x, y, z, t)) − d ((

∂2

∂x2 G(x, y, z, t)) + (
∂2

∂y2 G(x, y, z, t))

+(
∂2

∂z2 G(x, y, z, t))) = Dirac(x) Dirac(y) Dirac(z) Dirac(t)

The Green’s function G(x, y, z, t) is equal to 0 up to the instant that the source
is switched on. The Laplace transform of the first time derivative of G will be
expressed in terms of G(x, y, z, 0), which is set equal to 0.

> G(x,y,z,0):=0:

A functional operator F is formed for calculating the 1-dimensional Fourier
transform of an expression e with respect to an arbitrary variable v, the result
being given in terms of a second variable k. A similar functional operator K
is introduced to perform the 1-dimensional inverse Fourier transform of e from
k-space back to v-space.

> F:=(e,v,k)->fourier(e,v,k): K:=(e,k,v)->invfourier(e,k,v):

Using F, a 3-dimensional Fourier transform of pde is taken in the spatial coordi-
nates x, y, and z. This is implemented by nesting 3 one-dimensional transforms,
the result being expressed in terms of kx , ky , and kz . A Laplace transform in
time is then performed, and the result factored.

> eq1:=factor(laplace(F(F(F(pde,x,kx),y,ky),z,kz),t,s));

eq1 := laplace(fourier(fourier(fourier(G(x, y, z, t), x, kx ), y, ky), z, kz ), t, s)

(d kz 2 + s + d kx 2 + d ky2) = 1
eq1 is then solved for the Laplace-Fourier transform of G(x, y, z, t).

> eq2:=solve(eq1,laplace(F(F(F(G(x,y,z,t),x,kx),y,ky),z,kz),t,s));

eq2 :=
1

s + d kz 2 + d ky2 + d kx 2

Applying the inverse Laplace transform to eq2 puts us back into t-space.
> eq3:=invlaplace(eq2,s,t);

eq3 := e(−(d kz2+d ky2+d kx2) t)

Finally, by performing three 1-dimensional inverse Fourier transforms on eq3 ,
the desired Green’s function solution G to Equation (6.10) is obtained. It is
necessary to assume that both d and t are greater than zero for the transforms
to be explicitly calculated.

> G:=K(K(K(eq3,kx,x),ky,y),kz,z) assuming d>0,t>0;

G :=
1
8

e(−
x2+y2+z2

4 t d )

t d π(3/2)
√

t d
G is expressed in terms of the radial distance r by making the algebraic substi-
tution x2 + y2 + z2 =r2.

> G:=algsubs(xˆ2+yˆ2+zˆ2=rˆ2,G);

G :=
1
8

e(− r2

4 t d )

t d π(3/2)
√

t d
As expected for a point source, G has spherical symmetry, depending only on
the radial distance. The above form of G is valid for t > 0, with G=0 for t < 0.
Taking the diffusion coefficient d=1, G is animated over the radial range r=0
to 2.5 and time interval t=0.1 to 3, the opening frame of the animation being
shown in Figure 6.7.

> animate(eval(G,d=1),r=0..2.5,t=0.1..3,frames=100,

numpoints=200,thickness=2,labels=["r","G"]);

0

G

0.7

1 r 2

Figure 6.7: Initial frame of animation of the infinite-medium Green’s function.

As discussed in Wallace [Wal84], Green’s functions can be similarly derived for
semi-infinite, and even finite media, with specified boundary conditions. The
Green’s functions can in turn be used to solve problems involving distributed
time-dependent sources with the same boundary conditions. The interested
reader is referred to Wallace’s text.

6.2.5 Our Field of Dreams

A man is not old until regrets take the place of dreams.
John Barrymore, American actor, (1882–1942)

One summer I worked, along with other students, at a uranium mine on the
Arctic circle to earn sufficient money to pay my college expenses for the next
year. The students and the permanent workers would often play baseball under
the midnight sun on a field which consisted of processed mine tailings. On and
around this “field of dreams”, the students would chat about what they planned
to do in life. From this field of dreams came future engineers, chemists, and at
least one physicist. This recipe is dedicated to those friends of long ago.

Radioactive gas is diffusing at a steady rate into the atmosphere from leveled
mine tailings. Treating the ground and atmosphere as semi-infinite media, the
X-axis is taken vertically upwards with the ground-atmosphere boundary at
X =0. Letting d be the diffusion constant and λ the decay rate of the radioactive
gas (only one species is considered), the relevant modified diffusion equation for
the gas concentration C and boundary condition (Fick’s law) are of the form

∂C(X, T )
∂T

= d
∂2C(X, T )

∂X2 − λC(X, T ), −d
∂C(0, T )

∂X
= k, (6.11)

with T the time and k a positive constant (units of kg/(m2· s)). The various
constants may be removed from the equations by setting t=λT , x=

√
λ/dX,

and c(x, t)=(
√

dλ/k) C(X, T ). Then Equation (6.11) becomes

∂c(x, t)
∂t

=
∂2c(x, t)

∂x2 − c(x, t),
∂c(0, t)

∂x
= −1. (6.12)

Assuming that the concentration of radioactive gas in the atmosphere is initially
zero, our goal is to determine the normalized concentration c(x, t) in the region
x > 0 at arbitrary time t > 0 and animate the result.

In addition to the integral transform package, the DEtools package is loaded.
Using the latter will enable us to seek exponential solutions to the Laplace
transformed PDE which are convenient for solving the problem.

> restart: with(inttrans): with(DEtools):

The normalized diffusion equation in (6.12) is entered in pde.
> pde:=diff(c(x,t),t)=diff(c(x,t),x,x)-c(x,t);

pde :=
∂

∂t
c(x, t) = (

∂2

∂x2 c(x, t)) − c(x, t)

The addtable command is used to simplify the notation. The Laplace trans-
form of c(x, t) will be given as F (x).

> addtable(laplace,c(x,t),F(x),t,s):

The initial normalized concentration, c(x, 0), of radioactive gas in the atmo-
sphere is set equal to zero, and the Laplace transform is applied to pde.

> c(x,0):=0: eq:=laplace(pde,t,s);

eq := sF (x) = (
d2

dx2 F (x)) − F (x)

The expsols command is used to obtain exponential solutions for F (x) in eq .
> sol:=expsols(eq,F(x));

sol := [e(
√

s+1 x), e(−√
s+1 x)]

As x → ∞, the concentration of radioactive gas in the atmosphere must go to
zero. Therefore the exponential solution with the minus sign (the second result
in sol) is selected and multiplied by an arbitrary constant B.

> sol2:=F(x)=B*sol[2];

sol2 := F (x) = B e(−√
s+1 x)

The boundary condition at x=0 in (6.12) is applied as follows. The right-hand
side of sol2 is differentiated with respect to x and evaluated at x = 0. This
result is equated to the Laplace transform of −1.

> bc:=eval(diff(rhs(sol2),x),x=0)=laplace(-1,t,s);

bc := −B
√

s + 1 = −1
s

The Laplace transformed boundary condition, bc, is solved for B and the form
of the transformed concentration, F (x), displayed.

> B:=solve(bc,B); sol2;

B :=
1√

s + 1 s
F (x) =

e(−√
s+1 x)

√
s + 1 s

To facilitate the calculation of the inverse transform, let’s make a variable
change, substituting s = y − 1 on the rhs of sol2 and multiplying the result
by e−t to keep the Bromwich integral (6.9) the same.

> F2:=exp(-t)*subs(s=y-1,rhs(sol2));

F2 :=
e(−t) e(−√

y x)

√
y (y − 1)

Taking the inverse Laplace transform of F2 and assuming that x > 0 produces
the concentration c, the answer given in terms of the complimentary error func-
tion, erfc(z) = 1 − erf(z).

> c:=invlaplace(F2,y,t) assuming x>0;

c := −1
2

erfc(
x + 2 t

2
√

t
) ex +

1
2

erfc(−−x + 2 t

2
√

t
) e(−x)

The concentration c is now animated over the time interval t = 0 to 10, 100
frames being taken.

> plots[animate](c,x=0..10,t=0..10,frames=100,thickness=2,

numpoints=200,color=blue,labels=["x","c"]);
Execute the recipe to see how the concentration of radioactive gas in the atmo-
sphere builds up from zero to the steady-state concentration, where the diffusion
of radioactive gas molecules is balanced by their radioactive decay.

6.3 Bromwich Integral and Contour Integration

In performing inverse Laplace transforms in the previous section, we bypassed
evaluating the Bromwich integral by simply using the invlaplace command.
This approach is equivalent to consulting tables of Laplace transforms and their
inverses in the pre-computer age. Occasionally, Maple may stumble, unless
carefully guided, in performing the inverse Laplace transform, so it is instructive
to look at a few examples of evaluating the Bromwich integral.

Recall that the inverse Laplace transform L−1(F (s)) = f(t) is given by,

f(t) =
1

2πi

∫ c+i ∞

c−i ∞
est F (s) ds, t > 0, and f(t) = 0, t < 0. (6.13)

The integration involved in determining f(t) for t > 0 is to be performed along
a line s=c + i y in the complex s-plane, c being chosen so that s=c lies to the
right of all the poles of F (s).

The central idea is to replace the integral in (6.13) with the closed con-
tour integral 1

2πi

∮
Γ est F (s) ds, and apply Cauchy’s residue theorem. The basic

Bromwich contour Γ is shown on the left of Figure 6.8, consisting of a vertical
“leg” A → B between s=c− i L and s=c+ i L, and a circular arc B → C → A,
centered on the origin O, of radius R.

Re(s)

R

cC

B

A

O

Im(s)

c+iL

c-i L

cut
E

DC

B

A

F

Figure 6.8: Left: Basic contour. Right: Contour modified with branch cut.

For L → ∞, the integral contribution A → B will be the Bromwich integral.
By Cauchy’s residue theorem, assuming that the circular arc (B → C → A)
contribution approaches 0 as R → ∞, the Bromwich integral will be equal to
the sum of the residues of est F (s) at the poles of F (s) inside Γ.

A sufficient condition for
∫

B→C→A
est F (s) ds → 0 as R → ∞ is that

|F (s)| < constant/Rk, with k > 0. This condition always holds if F (s) =
P (s)/Q(s), where P (s) and Q(s) are polynomials and the degree of P is less
than that of Q. The sufficiency condition still holds if F (s) has branch points.

However, then the basic Bromwich contour must be modified. For example,
if F (s) has a branch point at the origin, the contour is modified as shown on the
right of Figure 6.8, a branch cut being introduced from −∞ to 0. The contour
is then given by A → B → C → D → E → F → A, the horizontal legs C → D
and E → F ultimately being shrunk onto the Re(s) axis and the small circular
arc D → E shrunk onto the origin.

6.3.1 Spiegel’s Transform Problem Revisited

When one pays a visit it is for the purpose of wasting
other people’s time, not one’s own.
Oscar Wilde, Anglo-Irish playwright, author, (1854–1900)

In Recipe 06-2-1, we successfully solved a fourth order inhomogeneous ODE
with specified initial conditions, suggested by Spiegel, using the laplace and
invlaplace commands in the integral transform package. In this recipe, we
shall solve the inverse Laplace transform of the previously obtained transform

F (s) = (s5 − 2 s4 + 6 s3 − 6 s2 + 5 s − 3)/(s6 + 3 s4 + 3 s2 + 1),
by evaluating the Bromwich integral using contour integration. In this case,
F (s) ≡ P (s)/Q(s) involves the ratio of two polynomials, the largest exponent of
s in P being one less than in Q. On the circular arc (radius R) of the Bromwich
contour Γ, s = R eiθ, so |F (s)| ∼ 1/R for sufficiently large R. Applying the
sufficiency condition, the circular arc contribution will be 0 for R → ∞. So,
taking the vertical leg to the right of all poles of F (s), the Bromwich integral
is equal to the sum of the residues of est F (s) at the poles of F (s) inside Γ.

F (s) is now entered and the Bromwich integrand est F (s) formed in f .
> restart:

> F(s):=(sˆ5-2*sˆ4+6*sˆ3-6*sˆ2+5*s-3)/(sˆ6+3*sˆ4+3*sˆ2+1);

F (s) :=
s5 − 2 s4 + 6 s3 − 6 s2 + 5 s − 3

s6 + 3 s4 + 3 s2 + 1
> f:=exp(s*t)*F(s);

f :=
e(s t) (s5 − 2 s4 + 6 s3 − 6 s2 + 5 s − 3)

s6 + 3 s4 + 3 s2 + 1
The denominator of f is factored in d, revealing that F (s) has third-order poles
at s2 + 1=0, i.e., at s = ± i.

> d:=factor(denom(f));

d := (s2 + 1)3

Setting the denominator of f equal to 0, and solving for s yields the explicit
locations (± I) of the poles, needed for evaluating the residues of F (s). Each
root is reproduced three times, again indicating that each pole is of third order.

> pole:=solve(denom(f)=0,s);

pole := −I, I, −I, I, −I, I

Creating a functional operator r to evaluate the residue at an arbitrary pole,
> r:=v->residue(f,s=pole[v]):

the residues of the two poles are determined in r1 and r2 .
> r1:=r(1); r2:=r(2);

r1 :=

1
8

I (8 t − 4 I t − 27
4

+ 8 I − t2

2
+

3
4

I (2 t − 16 + 5 I))

e(t I)

r2 :=
1
8

I(−8 t e(t I) − 4 I t e(t I) + 8 I e(t I) +
1
2

t2 e(t I) +
27
4

e(t I)

− 3
4

I e(t I) (−2 t + 16 + 5 I))

The inverse Laplace transform, ILT , (Bromwich integral), then is equal to the
sum of the two residues. Applying the complex evaluation command to the
sum and simplifying yields the following result:

> ILT:=simplify(evalc(r1+r2));

ILT := cos(t) +
5
8

t cos(t) − 21
8

sin(t) − 1
8

t2 sin(t) + 2 t sin(t)

The answer is further simplified by collecting sine and cosine terms.
> ILT:=collect(%,[sin,cos]);

ILT := (2 t − 21
8

− 1
8

t2) sin(t) + (1 +
5 t

8
) cos(t)

As a check, the inverse Laplace transform of F (s)=f/est is calculated with the
invlaplace command, yielding the same analytic result.

> check:=inttrans[invlaplace](f/exp(s*t),s,t);

check := −1
8

sin(t) (−16 t + t2 + 21) +
1
8

cos(t) (8 + 5 t)

6.3.2 Ms. Curious’s Branch Point

Propaganda is that branch of the art of lying which consists in nearly
deceiving your friends without quite deceiving your enemies.
F. M. Cornford, British author, poet, (1874–1943)

Ms. I. M. Curious has reached an important branch point in her educational
career. Although, she has done very well to this point in her course work,
she has to decide on whether to continue as a physics major and eventually
seek an academic career, or would she better off to switch into the engineering
physics program and go into industry. On seeking Jennifer’s advice, Jennifer
has temporarily deflected I. M.’s question by suggesting that, whatever decision
I. M. makes, this week’s homework is due tomorrow and no extensions will be
allowed. Ironically, one of the questions on this week’s quiz also deals with a
branch point, albeit of the mathematical variety.

Specifically, I. M. is requested to calculate L−1(e−a
√

s/s), with a > 0, using
contour integration. Then, employing this answer, determine L−1(e−a

√
s/

√
s).

Here is the recipe that I. M. has created. She begins by entering F (s) =
e−a

√
s/s, which has a branch point at s = 0, and then forming the Bromwich

integrand f =est F (s)/(2πi).
> restart:

> F(s):=exp(-a*sqrt(s))/s; f:=exp(s*t)*F(s)/(2*Pi*I);

F(s) :=
e(−a

√
s)

s
f :=

−1
2

I e(s t) e(−a
√

s)

s π
Choosing the modified contour Γ on the right of Figure 6.8, I. M. notes that the
only singular point in F (s) is at s=0, which is outside the contour. Therefore,
by Cauchy’s theorem, the closed line integral

∮
Γ f ds=0.

I. M. takes the radius of the outer circular arc (B → C and F → A in
Figure 6.8) to be R and the radius of the inner arc (D → E) to be ε. The limits
R → ∞ and ε → 0 are to be taken. As R → ∞, the outer arc contribution
(
∫

B→C
+
∫

F→A
) f ds → 0, and

∫
A→B

f ds is just the Bromwich integral, BI .
Then, BI =−(

∫
C→D

+
∫

D→E
+
∫

E→F
) f ds, the three integrals to be evaluated.

To determine IDE =
∫

D→E
f ds, I. M. sets s = s1 = ε eiθ. To evaluate

ICD =
∫

C→D
f ds, I. M. notes that θ=π for this leg, so that s=r eiπ =−r and

s2 ≡ √
s=

√
r eiπ/2 = i

√
r. For IEF =

∫
E→F

f ds, θ=−π so that s=r e−iπ =−r

still, but s3 ≡ √
s=

√
r e−iπ/2 =−i

√
r. I. M. now enters s1 , s2 , and s3 .,

> s1:=epsilon*exp(I*theta); s2:=sqrt(r)*exp(I*Pi/2);

s3:=sqrt(r)*exp(-I*Pi/2);

s1 := ε e(θ I) s2 :=
√

r I s3 := −I
√

r

The integral IDE is given by limε→0
∫ −π

π
(f i s)|s=s1 dθ, which is now evaluated.

> IDE:=int(limit(subs(s=s1,f*I*s),epsilon=0),theta=Pi..-Pi);

IDE := −1
To evaluate ICD and IEF , a functional operator G is introduced to calculate∫∞
0 (f)|√s=v,s=−r dr, where v is equal to s2 for ICD , and s3 for IEF . The inert

form of the integral command is used here, as Maple is not able to actually
perform the integrations.

> G:=v->Int(subs({sqrt(s)=v,s=-r},f),r=0...infinity):
Using G, ICD and IEF take the following forms. For the latter, I. M. inserts a
minus sign, since the integral IEF is in the opposite direction to ICD .

> ICD:= G(s2); IEF:=-G(s3);

ICD :=
∫ ∞

0

1
2

I e(−r t) e(−I a
√

r)

r π
dr IEF := −

∫ ∞

0

1
2

I e(−r t) e(a
√

r I)

r π
dr

The Bromwich integral BI then is equal to −(IDE + ICD + IEF ). This sum is
now converted to a trig form and the combine command applied.

> BI:=-combine(convert(IDE+ICD+IEF,trig));

BI := −
∫ ∞

0

cosh(r t) sin(a
√

r) − sinh(r t) sin(a
√

r)
r π

dr + 1

The integrand of BI can be simplified by making the algebraic substitution
cosh(rt) − sinh(rt)=e−rt.

> BI:=algsubs(cosh(r*t)-sinh(r*t)=exp(-r*t),BI);

BI := −
∫ ∞

0

sin(a
√

r) e(−r t)

r π
dr + 1

Applying the value command to BI yields the answer to the first part of the
question, the result being expressed in terms of the error (erf) function or,
on applying convert(BI,erfc), in terms of the complimentary error (erfc)
function.

> BI:=value(BI); BI:=convert(BI,erfc);

BI := −erf(
a

2
√

t
) + 1 BI := erfc(

a

2
√

t
)

So, L−1(e−a
√

s/s) = erfc(a/(2
√

t)).
As a check, I. M. will derive exactly the same result by directly applying

the inverse Laplace transform command. She mentally replaces a
√

s with
√

y
and t with t/a2 in the Bromwich integral, so that the function to be inverse
transformed is F2 = e−√

y/y. Loading the integral transform package,

> F2:=exp(-sqrt(y))/y; with(inttrans):

F2 :=
e(−√

y)

y
she inverse transforms F2 and applies the radical simplification command.

> check:=radsimp(invlaplace(F2,y,t/aˆ2));

check := erfc(
a

2
√

t
)

The result is exactly the same as in BI .
To answer the second part of the question, I. M. notes that e−a

√
s/

√
s =

− d
da (e−a

√
s/s) = −dF (s)

da . Thus, the second inverse Laplace transform, ILT2 ,
follows on differentiating BI with respect to a and multiplying by −1.

> ILT2:=-diff(BI,a);

ILT2 :=
e(−a2

4 t )√
π

√
t

Once again I. M. confirms her answer by making the same variable transforma-
tion as above, forming the new function F3 = e−√

y/(a
√

y), and applying the
inverse Laplace transform to F3 and simplifying.

> F3:=exp(-sqrt(y))/(a*sqrt(y));

F3 :=
e(−√

y)

a
√

y

> check2:=radsimp(invlaplace(F3,y,t/aˆ2));

check2 :=
e(−a2

4 t )√
π t

The answer, of course, is the same as in ILT2 .

6.3.3 Cooling That Weenie Rod

Write while the heat is in you. The writer who postpones the record-
ing of his thoughts uses an iron which has cooled to burn a hole with.
He cannot inflame the minds of his audience.
Henry David Thoreau, American author, philosopher, naturalist (1817–62)

Returning to our “tale” of the Northern weenie roast, having finished with
the long iron rods that they used for cooking their weenies, Russell places them
outdoors again to cool in the 0◦ temperature. This recipe is an idealization of
that cooling process.

A very long circular rod of radius R, initially having a constant temperature
T0 throughout, has a constant temperature of 0◦ applied to its surface for times
t ≥ 0. Determine the temperature at an arbitrary point inside the rod after the
cooling process has begun. Animate the temperature profile inside the rod.

If the rod were of finite length, one would use cylindrical coordinates (r, θ, z)
where r is the radial distance from the cylinder axis, θ the angle around the cir-
cumference of the cylinder, and z the distance along the cylinder axis. Treating
the rod as being infinitely long (which might have presented a “slight” problem
for the weenie roast!) removes the z-dependence from the problem. Since the
rod initially has a constant temperature throughout and the boundary condi-
tion at the surface has no angular dependence, the temperature T inside the
rod depends only on the radial distance r, i.e., T =T (r, t). T (r, t) then satisfies
the following heat diffusion equation (a2 being the heat diffusion coefficient) for
0 < r < R and t > 0,

∂T

∂t
= a2

(
∂2T

∂r2 +
1
r

∂T

∂r

)
, T (R, t) = 0, T (r, 0) = T0. (6.14)

The problem can be made dimensionless by setting u ≡ T/T0, x ≡ r/R, and
τ ≡ a2t/R2, so that

∂u

∂τ
=

∂2u

∂x2 +
1
x

∂u

∂x
, u(1, τ) = 0, u(x, 0) = 1. (6.15)

After loading the plots and integral transform packages, the normalized heat
flow equation given in (6.15) is entered in pde.

> restart: with(plots): with(inttrans):

> pde:=diff(u(x,tau),tau)=diff(u(x,tau),x,x)+diff(u(x,tau),x)/x;

pde :=
∂

∂τ
u(x, τ) = (

∂2

∂x2 u(x, τ)) +

∂

∂x
u(x, τ)

x
For later notational convenience, the addtable command is used to replace the
Laplace transform of u(x, τ), with respect to τ , with the symbol F (x, s).

> addtable(laplace,u(x,tau),F(x,s),tau,s):

The initial condition u(x, 0) = 1 is entered, and pde Laplace transformed with
respect to τ in de. The initial condition is automatically substituted into de.

> u(x,0):=1; de:=laplace(pde,tau,s);

u(x, 0) := 1

de := sF (x, s) − 1 = (
∂2

∂x2 F (x, s)) +

∂

∂x
F (x, s)

x
The spatially dependent ODE de is analytically solved for F (x, s) in sol , the
answer involving zeroth-order Bessel functions of the first and second kinds.

> sol:=dsolve(de,F(x,s));

sol := F (x, s) = BesselJ(0,
√−s x) F2 (s) + BesselY(0,

√−s x) F1 (s) +
1
s

Since Y0(
√−s x) diverges at x = 0 (the central axis of the rod), it is removed

from the right-hand side of sol , F (x, s) then taking the form shown in sol2 .
> sol2:=remove(has,rhs(sol),BesselY);

sol2 := BesselJ(0,
√−s x) F2 (s) +

1
s

From the boundary condition u(1, τ)=0, one must have F (x = 1, s)=0. This
transformed boundary condition (bc) is now applied to sol2 .

> bc:=eval(sol2,x=1)=0;

bc := BesselJ(0,
√−s) F2 (s) +

1
s

= 0

The boundary condition bc is then solved for the arbitrary function F2 (s)
which, because it is assigned, is automatically substituted into sol2 , or F (x, s).

> _F2(s):=solve(bc,_F2(s)); F(x,s):=sol2;

F2 (s) := − 1
BesselJ(0,

√−s) s

F (x, s) := −BesselJ(0,
√−s x)

BesselJ(0,
√−s) s

+
1
s

To determine u(x, τ), we must calculate the inverse Laplace transform of F (x, s).
As you may verify, using the inverse Laplace transform command in the inte-
gral transform package will not work here because, in addition to the simple
pole at s=0, F (x, s) has an infinite number of simple poles associated with the
zeros of J0(

√−s). This implies that u(x, τ) will consist of an infinite series. To
determine the series form, the Bromwich integral must be evaluated.

Choosing the “basic” contour Γ on the left of Figure 6.8, all the poles lie
inside Γ in the limit that the radius R of the circular arc goes to infinity. Since
the circular arc contribution to the line integral vanishes in this limit, the inverse
Laplace transform will be the sum of the residues of the poles inside Γ. We will
now calculate the residues.

For the pole at s=0, applying the residue command to esτF (x, s) yields 0.

> residue(exp(s*tau)*F(x,s),s=0);

0
Now the term J0(

√−s) in the denominator of the first term of F (x, s) has
simple zeros at, say, λ1, λ2,....Thus the integrand corresponding to the first
term has simple poles at s=sn = −λ2

n, n=1, 2, ... The first 15 values of sn are
calculated in S using the BesselJZeros command, and S is assigned.

> S:=[seq(s[n]=-(BesselJZeros(0,n))ˆ2,n=1..15)]; assign(S):

S := [s1 = −BesselJZeros(0, 1)2, s2 = −BesselJZeros(0, 2)2, · · ·
Care must be taken in evaluating the residues at the zeros of J0(

√−s). The
residue of the integrand at s=sn is given by

lim
s→sn

(s − sn)
(

−es τ J0(
√−s x)

s J0(
√−s)

)
= −

(
es τ J0(

√−s x)
s d

ds

(
J0(

√−s)
)
)

s=sn

,

where use has been made of L’Hospital’s rule. A functional operator U is
created to implement this latter form for the residue of the integrand at an
arbitrary zero of J0.

> U:=sn->simplify(-eval(exp(s*tau)*BesselJ(0,sqrt(-s)*x)

/(s*diff(BesselJ(0,sqrt(-s)),s)),s=sn));

U := sn → simplify

(
−e(s τ) BesselJ(0,

√−s x)
s ( d

ds BesselJ(0,
√−s)) s = sn

)

Making use of U , the formal series solution for u is given out to 15 terms.
> u:=Sum(U(s[n]),n=1..15);

u :=
15∑

n=1

(
2 e(sn τ) BesselJ(0,

√−sn x)
BesselJ(1,

√−sn)
√−sn

)
The explicit series form of the normalized temperature distribution is now de-
termined in u2 , only the first few terms being displayed here in the text.

> u2:=evalf(value(u));

u2 := 1.601974697 e(−5.783185964 τ) BesselJ(0., 2.404825558 x)

− 1.064799259 e(−30.47126234 τ) BesselJ(0., 5.520078110 x)

+ 0.8513991928 e(−74.88700679 τ) BesselJ(0., 8.653727913 x)

− 0.7296452400 e(−139.0402844 τ) BesselJ(0., 11.79153444 x) + · · ·
For τ > 0, the terms quickly drop off in magnitude, but for τ = 0 many terms
must be kept to reduce the “ringing” associated with the initial step-function
temperature profile.

The series solution u2 is animated over the time interval τ = 0.001 to 2. The
animation is started slightly past τ = 0 to eliminate the worst of the ringing.
The initial frame of the animation is shown in Figure 6.9.

> animate(u2,x=0..1,tau=0.001..2,numpoints=100,frames=50,

labels=["x=r/R","u=T/T0"]);

0

0.2

0.4

u=T/T0

0.8

1

0.2 0.4 x=r/R 0.8 1

Figure 6.9: Initial frame of animated temperature distribution inside the rod.

By the time the animation reaches τ = 2, the normalized temperature profile
inside the rod has effectively decayed to 0.

6.4 Other Transforms

While the Fourier and Laplace transforms are the most recognized and useful
of integral transforms, there exist other transforms which can be applied in
particular geometries. We will conclude this chapter with an example involving
the Hankel transform applied to a problem having cylindrical symmetry.

The Hankel transform (Fm(k)) of order m > −1/2 of a function f(r), and
the inverse transform, are given by

Fm(k) =
∫ ∞

0
f(r) Jm(kr) r dr, f(r) =

∫ ∞

0
Fm(k) Jm(kr) k dk, (6.16)

where Jm is the mth order Bessel function of the first kind. The reader is re-
ferred to standard mathematical physics texts for a discussion of the properties
of Hankel transforms. However, it should be mentioned that there is no simple
convolution theorem for the Hankel transform.

The Maple commands for performing the transform operations in equa-
tion (6.16) are hankel(f(r),r,k,m) and hankel(Fm(k),k,r,m), respectively.
The integral transform package must be loaded to use these commands. In
some cases, as the following recipe illustrates, it may be easier to make use of
the basic defining relations (6.16) than Maple’s Hankel transform commands.

6.4.1 Meet the Hankel Transform

Reason transformed into prejudice is the worst form of prejudice,
because reason is the only instrument for liberation from prejudice.
Allan Bloom, American educator, author, (1930–1992)

We wish to determine the subsequent transverse vibrations of a thin, infi-
nite, elastic membrane which is initially at rest and has an initial displacement
ψ(r, θ, t = 0)=A/

√
1 + r2/a2. The solution ψ(r, θ, t > 0) will then be animated

in three dimensions, taking A=1, a=10, and wave speed c=1.
The plots, integral transform, and VectorCalculus packages are loaded, being

required for the animate, hankel, and Laplacian commands, respectively.
> restart: with(plots): with(inttrans): with(VectorCalculus):

It is assumed that c > 0, t > 0, A > 0, a > 0, and r > 0. The initial condition
for the shape is also entered.

> assume(c>0,t>0,A>0,a>0,r>0): ic:= A/sqrt(1+rˆ2/aˆ2);

ic :=
A√

r2

a2 + 1

Since the initial shape has no angular (θ) dependence, the subsequent membrane
displacement must also be independent of θ, i.e., ψ = ψ(r, t). This also suggests
that we set m = 0 when applying the Hankel transform and its inverse. The
relevant wave equation for ψ(r, t) is now entered in polar coordinates in pde.

> pde:=expand(Laplacian(psi(r,t),’polar’[r,theta]))

-(1/cˆ2)*diff(psi(r,t),t,t)=0;

pde :=

∂

∂r
ψ(r, t)

r
+ (

∂2

∂r2 ψ(r, t)) −
∂2

∂t2
ψ(r, t)

c2 = 0

Then pde is Hankel transformed with respect to the radial coordinate r, with
m=0, and multiplied by −c2. A time-dependent ODE results in eq .

> eq:=-cˆ2*hankel(pde,r,k,0);

eq := (
∂2

∂t2
hankel(ψ(r, t), r, k, 0)) + k2 hankel(ψ(r, t), r, k, 0) c2 = 0

To simplify the notation, the Hankel transform of ψ(r, t) with respect to r with
m=0 is replaced with F (t) in eq .

> eq2:=subs(hankel(psi(r,t),r,k,0)=F(t),eq);

eq2 := (
d2

dt2
F (t)) + k2 F (t) c2 = 0

Since ∂ψ(r, t)/∂t=0 at t=0, eq2 is analytically solved using dsolve for F (t),
subject to the initial condition dF (t)/dt=0 at t=0.

> sol:=dsolve({eq2,D(F)(0)=0},F(t));
sol := F(t) = C2 cos(c k t)

At t = 0, one has from eq2 , C2 = F (0). So we need to calculate the Hankel
transform (with m=0) of the initial membrane shape. If the hankel command
was used, C2 would be expressed as a combination of generalized hypergeomet-
ric functions, which are difficult to reduce to a simpler form. This then would
lead to further difficulty in performing the inverse Hankel transform. Instead,
let’s determine C2 by using the basic defining relation (6.16) and calculating∫∞
0 ic r J0(kr) dr. The result is then simplified.

> _C2:=simplify(int(ic*r*BesselJ(0,k*r),r=0..infinity));

C2 :=
A e(−a k) a

k
With the form of C2 automatically substituted, F (t) is given by the rhs of sol .

> F(t):=rhs(sol);

F (t) :=
A e(−a k) a cos(c k t)

k
Applying the inverse Hankel transform command to F (t) would again generate
an answer in terms of generalized hypergeometric functions which is difficult to
simplify and express in real form. So, the basic defining relations will be used.
First, however, it’s necessary to convert F (t) to an exponential form.

> F2:=simplify(convert(F(t),exp));

F2 :=
1
2

A a (e(k (−a+c t I)) + e(−k (a+c t I)))
k

Then ψ is obtained by calculating the integral
∫∞
0 F2 k J0(kr) dk.

> psi:=int(F2*k*BesselJ(0,k*r),k=0..infinity);

ψ :=
A a

2 r

√
1 +

(a − c t I)2

r2

+
A a

2 r

√
1 +

(a + c t I)2

r2

The displacement ψ is expressed in terms of I ≡ √−1. As the initial profile was
real, ψ should be real, however, not complex. That it is real may be confirmed
by applying the complex evaluation command to ψ and simplifying.

> psi2:=simplify(evalc(psi));

ψ2:= Aa
√

2
√√

r4+2 r2 a2−2 r2 c2 t2+a4+2 a2 c2 t2+c4 t4+r2+a2−c2 t2/
(2

√
r4+2 r2 a2−2 r2 c2 t2+a4+2 a2 c2 t2+c4 t4)

To animate the vibrations, the given parameter values are substituted into ψ2,
which is also converted to Cartesian coordinates by setting r=

√
x2 + y2.

> psi3:=subs({A=1,a=10,c=1,r=sqrt(xˆ2+yˆ2)},psi2);
Finally, ψ3 is animated with the animate command. A triangular grid style is
used to smooth out the wave form.

> animate(plot3d,[psi3,x=-100..100,y=-100..100],t=0..90,
frames=100,axes=boxed,gridstyle=triangular,orientation
=[25,65],tickmarks=[3,3,2],labels=["x","y","psi"]);

Execute the recipe to see what happens.

6.5 Supplementary Recipes
06-S01: Verifying the Convolution Theorem
Using the symmetric factor convention, verify that the convolution theorem is
satisfied for the functions f1 =A e−a2x2

and f2 =A e−a |x| with a > 0, A > 0.

06-S02: Bandwidth Theorem
An approximately monochromatic plane wave packet in one dimension has the
instantaneous form u(x, 0) = f(x) eik0x, with f(x) the envelope function and
k0 the central wave number. Consider:

(a) f1 =A e−a2x2
; (b) f2 =A e−a|x|; (c) f3 =A for |x| < a, 0 for |x| > a.

For each of the f(x), explicitly evaluate the root mean square deviations from
the means, i.e., ∆x =

√
< x2 > − < x >2 and ∆k =

√
< k2 > − < k >2. The

means < > are defined in terms of the respective intensities |u(x, 0)|2 = |f(x)|2
and |F (k)|2, respectively, where F (k) is the Fourier transform of f(x). Show
that in each case the bandwidth theorem (the optical analogue of the uncertainty
principle) ∆x∆k ≥ 1/2 is satisfied.

06-S03: Solving an Integral Equation
Solve the following integral equation for f(x), and plot the solution:∫ ∞

0
f(x) cos(α x) dx = {1 − α for 0 ≤ α ≤ 1, 0 for α > 1}.

06-S04: Verifying Parseval’s Theorem
For f(x)=x6 sin(x) e−2x2

, calculate the Fourier transform F (k)=F(f(x)) using
the symmetric convention. Plot the intensities |f(x)|2 and |F (k)|2. Verify that
Parseval’s theorem is satisfied.

06-S05: Heat Diffusion in a Copper Rod
Suppose that the initial temperature distribution inside an infinitely long, thin,
insulated, copper rod is T = T0 for −x0 ≤ x ≤ x0 and zero otherwise. By
Fourier transforming the heat diffusion equation with respect to x, solving the
resulting time-dependent ODE with the initial condition, and performing the
inverse transform, determine the temperature distribution inside the rod for
arbitrary time t > 0. For copper, the thermal conductivity K =386 W/m·K◦,
the density ρ = 8900 kg/m3, and the specific heat capacity C = 390 J/kg· K◦.
If x0 = 1 m and T0 = 100 ◦, animate the temperature profile over the time
interval t=0 to the time it takes for the temperature at the origin to drop to
50 ◦. Determine the temperature at x=2 m at the time the temperature at the
origin has dropped to 50◦.

06-S06: Solving Another Integral Equation
Using the convolution theorem, solve the following integral equation for y(x):∫ ∞

−∞
(y(u)/((x − u)2 + a2)) du = 1/(x2 + b2)4, 0 < a < b.

Verify the solution by direct substitution of y(u) back into the integral equation.
Taking a=1 and b=2, plot y(x) over the range x=−5 to +5.

06-S07: Free Vibrations of an Infinite Beam
Consider a horizontal beam which is initially at rest (∂ψ(x,0)/∂t=0) with the
shape ψ(x,0)=A e−b2x2

, A and b being positive constants. When released, the
transverse vibrations of the beam are governed by ψxx + (1/a2) ψtt = 0, with
a positive and the subscripts denoting partial derivatives. By Fourier trans-
forming the beam PDE with respect to x, solving the resulting time-dependent
ODE with the initial conditions, and performing the inverse transform, deter-
mine ψ(x, t) for t > 0. Taking A=b=a=1, animate the motion of the beam.

06-S08: A Potential Problem
The potential V (x, y) (in volts) in the region x > 0, y > 0 has the boundary
conditions V (0, y)=0 and V (x, 0)=1. Using an appropriate Fourier transform
approach, determine V (x, y). Plot the equipotentials in 0.1 volt increments.

06-S09: Solving an ODE
Use the Laplace transform method to solve the ODE

y ′′′(t) + 8 y(t) = 32 t3 − 16 t, y(0) = y ′(0) = y ′′(0) = 0.
Verify the solution by directly solving the ODE using the dsolve command
with the Laplace transform option. Plot y(t) over the range t = 0 to 1.6, and
determine qualitatively and quantitatively the time at which the first minimum
in the solution curve occurs.

06-S10: Impulsive Force
A unit mass attached to a spring and lying on a flat rough table experiences
an impulsive force F (t)=5 t3 sin(t) for t ≤ π and = 0 for t ≥ π, the ODE being

y ′′(t) + 2 y ′(t) + y(t) = F (t), y(0) = 0, y ′ = 1.
Use the Laplace transform method to determine the displacement y(t) of the
mass. Confirm your result with the dsolve command, using the Laplace trans-
form option. Plot the solution over the range t=0 to 10. Determine the maxi-
mum displacement and the time at which it occurs.

06-S11: Bromwich Integral
Consider F (s) = 1/(s6 − 6s5 + 9s4 + 8s3 − 24s2 + 16). Identify the singular
points of F (s). Calculate L−1(F (s)) by replacing the Bromwich integral with an
appropriate closed contour integral and using Cauchy’s residue theorem. Verify
the inverse Laplace transform by directly using the invlaplace command.

06-S12: Branch Point
Consider F (s) =

√
s/(s − 1). Identify the singular points. Plot a modified

Bromwich contour which can be used to evaluate L−1(F (s)). Use this contour
to explicitly carry out the evaluation. Verify your answer by directly using the
invlaplace command.



Chapter 7

Calculus of Variations
Although one could solve variational calculus problems with computer algebra
by completely mimicking a hand calculation, in all the recipes of this chapter
we will use the VariationalCalculus package to ease the work.

7.1 Euler–Lagrange Equation

Among all functions y(x) with fixed points y(x0) = y0 and y(x1) = y1 at two
distinct points A and B respectively, find the y(x) which gives an extremum
(minimum or maximum) or stationary value to the integral

I[y] =
∫ x1

x0

F (x, y, y ′) dx. (7.1)

The value of the integral I depends on what functional form is chosen for y(x).
In a given problem, F is a known function of x, y, and y ′ ≡ dy

dx . The y which
gives an extremum value to I is found by solving the Euler–Lagrange equation,

∂F

∂y
− d

dx

(
∂F

∂y ′

)
= 0. (7.2)

In physical problems, it is usually evident whether y minimizes or maximizes I.
If F does not explicitly depend on x, Equation (7.2) may be written as

d

dx

(
F − y ′ ∂F

∂y ′

)
= 0, (7.3)

so that a first integral of Equation (7.2) is F − y ′(∂F/∂y ′) = constant.

7.1.1 Betsy’s In A Hurry

Work expands so as to fill the time available for its completion.
C. Northcote Parkinson, English writer, Parkinson’s Law (1958)

One of the oldest variational examples is the brachistochrone1 problem pro-
posed by John Bernoulli in 1696 and independently solved by him and his
brother James, Gottfried Leibniz, and Isaac Newton.

1brachistos ≡ shortest, chronos ≡ time

Consider a tiny bug (Betsy Bug) of mass m starting from rest and sliding
under the influence of gravity along a smooth greased wire from some fixed
point A to another fixed point B somewhere (but not directly) below A. What
should the shape y(x) of the wire be between A and B so that Betsy’s time of
descent is a minimum?

Let’s take the point A to be at the origin (i.e., x0 =0, y0 =0) and measure
y downwards. Neglecting friction and air resistance, Betsy’s speed v when she
passes through an arbitrary lower point P (x, y) is obtained by equating her in-
crease in kinetic energy to the decrease in potential energy, i.e., 1

2 m v2 = m g y,
or v =

√
2gy, where g is the acceleration due to gravity. But v = ds/dt where

ds=
√

(dx)2 + (dy)2 =
√

1 + (y ′)2 dx is an element of arc length along the path
at P and dt is the time it takes for Betsy to traverse ds. Combining these
results, the time T of descent from A to B(where x = x1, y = y1) is

T =
∫ x1

0

√
1 + (y ′)2 dx√

2gy
.

The y(x) which minimizes T follows on solving (7.2) with F =
√

1 + (y ′)2/
√

2gy.
In this recipe, we will derive Betsy’s path in the parametric form

x =
a

2
(θ − sin θ), y =

a

2
(1 − cos θ),

with θ a parameter which is equal to 0 at A. Then, taking θ=θ1 at B, we will
show that the minimum time of descent is T =

√
a/2g θ1 . Finally, choosing

x1 = 5 m, y1 = 2 m, Betsy’s path will be plotted and T evaluated.
If the form of F is specified, the EulerLagrange command will calculate the

left-hand side of (7.2), and even produce the first integral if possible. To use
this command, the VariationalCalculus package must be first loaded.

> restart: with(VariationalCalculus):

The relevant F for the brachistochrone problem is entered. The factor 1/
√

2g
is omitted since it will cancel out of the Euler–Lagrange equation, (7.2).

> F:=sqrt(1+diff(y(x),x)ˆ2)/sqrt(y(x));

F :=

√
1 + (

d

dx
y(x))2√

y(x)
The EulerLagrange command, with F , x, and y(x) given as arguments, is
applied to F and simplified, generating two results.

> eq:=simplify(EulerLagrange(F,x,y(x)));

eq :=

⎧⎪⎨
⎪⎩

1√
1+(

d

dx
y(x))2

√
y(x)

= K1,−1
2

1+(
d

dx
y(x))2+2 (

d2

dx2 y(x)) y(x)

y(x)(3/2) (1+(
d

dx
y(x))2)(3/2)

⎫⎪⎬
⎪⎭

Since F doesn’t explicitly depend on x here, the first integral is generated in
the first expression in the above output, the integration constant being K1.
The second expression is just the left-hand side of the resulting ODE generated

by the Euler–Lagrange equation. We will work with the first integral and now
select it in eq2 . The last argument, [1], in the command line removes the curly
(Maple set) brackets which would otherwise enclose the first integral expression.

> eq2:=select(has,eq,K[1])[1];

eq2 :=
1√

1 + (
d

dx
y(x))2

√
y(x)

= K1

The dsolve command is used to analytically solve eq2 for y(x). The parametric
option is specified, since a parametric form of the solution is desired.

> sol:=dsolve(eq2,y(x),parametric);

sol :=

[
y( T ) =

1
K1

2 (1 + T 2)
,

x ( T ) =
− T − arctan( T ) − arctan( T ) T 2 + C1 K1

2 + C1 K1
2 T 2

K1
2 (1 + T 2)

]

The mathematical forms of x and y are given in terms of the parameter T . The
quantity C1 is a second integration constant. A new parameter θ is introduced
by setting T =cos(θ/2)/ sin(θ/2). The constant K1 is set equal to 1/

√
a.

> _T:=cos(theta/2)/sin(theta/2): K[1]:=1/sqrt(a):

Let’s tackle x first. The right-hand side of the second expression in sol is
symbolically simplified with the trig option, the above assignments having been
automatically substituted.

> x:=simplify(rhs(sol[2]),trig,symbolic);

x := −cos(
θ

2
) a sin(

θ

2
) − arctan

⎛
⎜⎝cos(

θ

2
)

sin(
θ

2
)

⎞
⎟⎠ a + C1

Applying the combine command with the trig option reduces the first term in
x to one of the terms desired in the final form of x.

> x:=combine(x,trig);

x := −1
2

a sin(θ) − arctan

⎛
⎜⎝cos(

θ

2
)

sin(
θ

2
)

⎞
⎟⎠ a + C1

We must choose the constant C1 in such a way that the last two terms in the
above output reduce to a θ/2. To do this, recall that the parameter θ must have
the value 0 when x=0. For θ=0, the first term in the above output is 0, while
the second term yields −a arctan(∞), or −a π/2. Thus, to make θ=0 at x=0,
we must choose C1 = a π/2. This substitution is now made in x. We further
substitute cos(θ/2)=cot(θ/2) sin(θ/2).

> x:=subs({_C1=a*Pi/2,cos(theta/2)=cot(theta/2)*sin(theta/2)},x);

x := −1
2

a sin(θ) − arctan(cot(
θ

2
)) a +

a π

2
Applying the combine command with the trig option, and then making the
substitution arccot(cot(θ/2))=θ/2 produces the desired final form for x.

> x:=combine(x,trig);

x := −1
2

a sin(θ) + a arccot(cot(
θ

2
))

> x:=subs(arccot(cot(theta/2))=theta/2,x);

x := −1
2

a sin(θ) +
a θ

2
Next, we tackle y by selecting the first expression from the right-hand side of
sol and simplifying with the trig and symbolic options.

> y:=simplify(rhs(sol[1]),trig,symbolic);

y := a sin(
θ

2
)2

Applying combine, with the trig option, yields the final desired form of y.
> y:=combine(y,trig);

y :=
a

2
− 1

2
a cos(θ)

With the parametric forms of x and y determined, the minimum time of descent
can be calculated. First note that the time of descent

T =
∫ x1

0

√
1 +
(

dy(x)
dx

)2

dx√
2 g y(x)

=
∫ θ1

0

√
1 +
(

dy(θ)/dθ

dx(θ)/dθ

)2 (
dx(θ)

dθ

)
dθ√

2 g y(θ)
,

where θ1 is the value of θ at the end point B. The latter integrand is now
calculated and simplified assuming that θ > 0.

> integrand:=simplify(sqrt(1+(diff(y,theta)/diff(x,theta))ˆ2)

*diff(x,theta)/sqrt(2*g*y)) assuming theta>0;

integrand :=
1
2

√
2 a√
g a

The minimum time of descent, assigned the name T , follows on integrating the
integrand from θ=0 to θ1 .

> T:=int(integrand,theta=0..theta1);

T :=
1
2

√
2 a θ1√
g a

The value of θ1 depends on the coordinates x1, y1 of the point B, and must be
determined numerically. This is now done for the specified values x1 =5, y1 =2.

> x1:=x=5; y1:=y=2;

x1 := −1
2

a sin(θ) +
a θ

2
= 5 y1 :=

a

2
− 1

2
a cos(θ) = 2

The above pair of equations is solved numerically for the constant a and the
value of θ at B, i.e., θ1 .

> sol2:=fsolve({x1,y1},{a,theta});
sol2 := {θ = 3.819665136, a = 2.248728127}

The values of θ1 and a are now expressed separately.
> theta1:=eval(theta,sol2); a:=eval(a,sol2);

θ1 := 3.819665136 a := 2.248728127
Betsy’s path then is given by the following expressions for x and y. Taking
g=9.8 m/s2, her time of descent is also calculated.

> x:=x; y:=y; g:=9.8: T:=evalf(T);

x := −1.124364064 sin(θ) + 1.124364064 θ

y := 1.124364064 − 1.124364064 cos(θ)

T := 1.293795782
Along the path given by the above forms of x and y, Betsy takes about 1.29
seconds to travel from the origin (A) to B. This is the minimum time possible
between these points, any other path producing a longer time. If you don’t
believe it, try calculating the time for descent along, e.g., a straight line path
between A and B.

Finally, Betsy’s path is plotted from θ = 0 to θ1 with constrained scaling.
The resulting picture is shown in Figure 7.1. Mathematically, the path is a
portion of an inverted cycloid. Another example of a cycloidal path is the
trajectory traced out by a point on the rim of a wheel rolling without slipping.

> plot([x,-y,theta=0..theta1],scaling=constrained,

thickness=2,labels=["x","y"]);

–2

y

–1

0
1 2 3 4 5

x

Figure 7.1: Betsy’s path which minimizes the time of descent.

7.1.2 Fermat’s Principle

The light that radiates from the great novels time can never dim....
Milan Kundera, Czech author, critic, (1929–)

Fermat’s principle states that a ray of light in a medium with a variable re-
fractive index n will follow the path which requires the shortest traveling time.
Consider a medium which has a refractive index n(x, y)=eay, with x the hori-
zontal and y the vertical coordinate, respectively. Noting that the speed of light
is v=c/n, where c is the vacuum speed of light, determine the general equation
for the light ray path. Taking a=1, determine the light ray path between the
points (−1, 1) and (1, 1) and plot it.

Now the light ray speed v = ds/dt = c/n, where ds =
√

(dx)2 + (dy)2 =√
1 + (y ′)2 dx is an element of arclength and dt is a time interval. So, since

n=eay, the total time to travel between x=−1 and +1 is given by

T =
1
c

∫ 1

−1
eay
√

1 + (y ′)2 dx.

Omitting the constant factor 1/c which cancels out, the Euler–Lagrange equa-
tion will be solved with F =eay

√
1 + (y ′)2. As in the previous recipe, since F

doesn’t explicitly depend on x, a first integral must exist.
After loading the VariationalCalculus package, the refractive index n is en-

tered and F formed.

> restart: with(VariationalCalculus):

> n:=exp(a*y(x)); F:=n*sqrt(1+diff(y(x),x)ˆ2);

n := e(a y(x))

F := e(a y(x))

√
1 + (

d

dx
y(x))2

The EulerLagrange command is applied to F and the result simplified.

> eq:=simplify(EulerLagrange(F,x,y(x)));

eq :=

⎧⎪⎪⎨
⎪⎪⎩

e(a y(x))√
1+(

d

dx
y(x))2

= K1, −
e(a y(x)) (−a − a (

d

dx
y(x))2+(

d2

dx2 y(x)))

(1+(
d

dx
y(x))2)(3/2)

⎫⎪⎪⎬
⎪⎪⎭

In addition to generating the left-hand side of the Euler–Lagrange equation (sec-
ond term in eq), the first integral has been generated (first term) as expected.
Once again, we will work with the first integral, selecting the expression from
eq which contains the constant K1.

> eq2:=select(has,eq,K[1])[1];

eq2 :=
e(a y(x))√

1 + (
d

dx
y(x))2

= K1

Then, eq2 is analytically solved for y(x) using the dsolve command and as-
suming that K1 is real. If this assumption is not made, a much more formidable
form of the solution will occur. Further, even with the assumption included,
there occasionally may appear more possible forms of y(x) in the output of sol .
If this occurs, either select the solution which corresponds to the one chosen
here or re-execute the work sheet to get an identical output.

> sol:=dsolve(eq2,y(x)) assuming K[1]::real;

sol := y(x) =
1
2

ln(K1
2)

a
, y(x) =

ln(

√
tan(a (x − C1 ))2 + 1K1

tan(a (x − C1 ))
)

a
,

y(x) =
ln(−

√
tan(a (x − C1 ))2 + 1K1

tan(a (x − C1 ))
)

a
Noting that the second and third solutions in sol turn out to be equivalent, the
rhs of the third solution is chosen and converted to sines and cosines.

> y:=convert(rhs(sol[3]),sincos);

y :=

ln

⎛
⎜⎜⎜⎜⎝−

√
sin(a (x − C1 ))2

cos(a (x − C1 ))2
+ 1 K1 cos(a (x − C1 ))

sin(a (x − C1 ))

⎞
⎟⎟⎟⎟⎠

a
The form of y is considerably simplified in y2 by applying the radical simplifi-
cation command followed by combine with the trig option.

> y2:=combine(radsimp(y),trig);

y2 :=
ln(− K1

sin(a x − a C1 )
)

a
With an analytic form for the light ray path determined, the given parameter
values are now entered. Here h is the vertical coordinate of the end points of
the light ray path and x=±L will be the end points’ horizontal coordinates.

> a:=1: h:=1: L:=1:

The light ray must pass through the points (x=L, y=h) and (−L, h). These
boundary conditions are entered in bc1 and bc2 , respectively.

> bc1:=eval(y2,x=L)=h; bc2:=eval(y2,x=-L)=h;

bc1 := ln(
K1

sin(−1 + C1 )
) = 1 bc2 := ln(

K1

sin(1 + C1 )
) = 1

Then bc1 and bc2 are numerically solved for K1 and C1 and the solution
assigned. K1 and C1 are automatically substituted into y2 which is displayed.

> sol2:=fsolve({bc1,bc2},{K[1],_C1}); assign(sol2): y2:=y2;

sol2 := { C1 = −7.853981634, K1 = −1.468693940}

y2 := ln(
1.468693940

sin(x + 7.853981634)
)

The solution y2 is now plotted with constrained scaling,
> plot(y2,x=-L..L,thickness=2,tickmarks=[3,2],view=

[-L..L,0..h],scaling=constrained,labels=["x","y"]);

0

1

y

–1 1x

Figure 7.2: Path traced by light ray between the points (−1, 1) and (1, 1).

the light ray path being shown in Figure 7.2. The path joining the two points
is curved rather than a straight line.

7.1.3 Betsy’s Other Path

Adulthood is the ever-shrinking period between childhood and old age.
It is the apparent aim of modern industrial societies to reduce this
period to a minimum.
Thomas Szasz, American psychiatrist, The Second Sin,“Social Relations” (1973)

Suppose that we want to find the function y(x) which minimizes or maximizes
I[y] =

∫ x1

x0
F (x, y, y ′) dx, with y fixed at x0, but with the other end point x1

free to lie anywhere along the curve G(x, y) = 0. According to Mathews and
Walker [MW71], y(x) may still be found by solving the Euler–Lagrange equa-
tion, but subject to the subsidiary condition(

F − y ′ ∂F

∂y ′

)
∂G

∂y
− ∂F

∂y ′
∂G

∂x
= 0. (7.4)

For the brachistochrone, F =
√

(1 + (y ′)2)/y and the condition (7.4) reduces
to y ′ = (∂G/∂y)/(∂G/∂x). I.e., the curve of quickest descent must intersect
G(x, y)=0 at right angles.

As a simple illustration of applying this latter condition, suppose that Betsy
bug now slides down a smooth wire from the origin (starting at rest, and with y
measured downwards) to the parabola y=m x2 + c, where the parabola passes

through the two points (x=0, y=50) and (30, 0) cm. Determine the path which
minimizes Betsy’s time of descent to the parabola. Taking g=980 cm/s2, how
long does it take Betsy to reach the parabola and through what distance does
she drop? Plot Betsy’s path and the parabola, using constrained scaling.

After loading the plots package, the parabolic equation G is entered.
> restart: with(plots):

> G:=Y-m*Xˆ2-c=0;

G := Y − m X2 − c = 0
Evaluating G at the two given points and solving yields the values of m and c.

> sol:=solve({eval(G,{X=0,Y=50}),eval(G,{X=30,Y=0})},{m,c});

sol := {c = 50, m =
−1
18

}
Assigning sol , the explicit form of the parabola G then is as follows.

> assign(sol): G:=G;

G := Y +
X2

18
− 50 = 0

The quantity (∂G/∂Y )/(∂G/∂X) is calculated and evaluated at the (still un-
known) end point X =x1.

> s:=eval(lhs(diff(G,Y))/lhs(diff(G,X)),X=x1);

s :=
9
x1

Since F is the same as in Recipe 07-1-1, the path of quickest descent must
have the same mathematical form as obtained earlier. In parametric form, the
equations describing the x and y coordinates of the path are as follows:

> x:=(a/2)*(theta-sin(theta)); y:=(a/2)*(1-cos(theta));

x :=
1
2

a (θ − sin(θ)) y :=
1
2

a (1 − cos(θ))

The slope of the path, dy/dx or (dy/dθ)/(dx/dθ), must be equal to s at some
unknown value θ1 of the parameter θ. This boundary condition is now entered.

> bc||1:=eval(diff(y,theta)/diff(x,theta),theta=theta1)=s;

bc1 :=
sin(θ1)

1 − cos(θ1)
=

9
x1

As two additional conditions, x and y evaluated at θ=θ1 must be equal to x1
and y1 . Finally, in bc4 we evaluate G at X =x1 and Y =y1 .

> bc||2:=x1=eval(x,theta=theta1);

bc2 := x1 =
1
2

a (θ1 − sin(θ1))

> bc||3:=y1=eval(y,theta=theta1);

bc3 := y1 =
1
2

a (1 − cos(θ1))

> bc||4:=eval(G,{X=x1,Y=y1});

bc4 := y1 +
x1 2

18
− 50 = 0

The 4 boundary conditions are numerically solved for x1 , y1 , a, and θ1.
> sol2:=fsolve({seq(bc||i,i=1..4)},{x1,y1,a,theta1},x1=0..30);

sol2 :={a= 26.49236117,x1 = 22.15176281,θ1= 2.369761726,y1 = 22.73885580}
Assigning sol2 , the parametric equations describing Betsy’s path are as follows:

> assign(sol2): x:=x; y:=y;

x := 13.24618058 θ − 13.24618058 sin(θ)

y := 13.24618058 − 13.24618058 cos(θ)
Entering the value of g, the distance h through which Betsy drops and the
(minimum) time T it takes to reach the parabola are calculated. h is given by
the numerical value of y1 , while for T we borrow the result T =

√
a/(2g) θ from

recipe 07-1-1 and evaluate it at θ=θ1.
> g:=980: h:=evalf(y1); T:=evalf(sqrt(a/(2*g))*theta1);

h := 22.73885580 T := 0.2755097538
Betsy drops through 22.7 cm and takes about 0.28 seconds to reach the parabola.
In gr1 and gr2, Betsy’s path and the parabola are plotted, respectively.

> gr1:=plot([x,-y,theta=0..theta1],color=blue,thickness=2):

> gr2:=plot(lhs(G)-Y,X=0..30,thickness=2,linestyle=3):

The two graphs are superimposed with the display command, constrained
scaling being used. The resulting picture is shown in Figure 7.3, the solid curve
being Betsy’s path, the dashed curve the parabola.

> display({gr1,gr2},scaling=constrained,labels=["x","y"]);

–50

0

y

15 30x

Figure 7.3: Betsy’s path which minimizes the time of descent to the parabola.

Since constrained scaling has been used, it may be seen from the figure that
Betsy’s path does indeed intersect the parabola at right angles.

7.2 Subsidiary Conditions

In differential calculus, we may require a function φ(x, y) to be a conditional
minimum or maximum subject to a subsidiary condition y = y1(x) or, equiva-
lently g(x, y)=0. How is a conditional extremum calculated?

A straightforward way is to replace y in φ(x, y) with y1(x) and calculate
dφ(x, y1(x))

dx
= 0, or using the chain rule,

∂φ

∂x
+

∂φ

∂y1

∂y1

∂x
= 0. (7.5)

An alternate procedure is the method of Lagrange multipliers. One extrem-
izes φ + λ g, subject to g(x, y) = 0. λ is called the Lagrange multiplier. More
explicitly, one must solve the simultaneous equations

∂

∂x
(φ + λ g) = 0,

∂

∂y
(φ + λ g) = 0, g = 0. (7.6)

In variational calculus, we may require an integral I[y] =
∫ x1

x0
F (x, y, y ′) dx

to be a minimum or maximum subject to an integral subsidiary condition
N =

∫ x1

x0
G(x, y, y ′) = C, where C is a known constant. In this case, one sets

F =F + λG and solves the Euler–Lagrange equation
∂F
∂y

− d

dx

(
∂F
∂y ′

)
= 0. (7.7)

Extensions can be made to handle more than one subsidiary condition. In the
following recipes, examples from differential and variational calculus are given.

7.2.1 Ground State Energy

Authority has always attracted the lowest elements in the human
race. All through history mankind has been bullied by scum...Every
government is a parliament of whores. The trouble is, in a democ-
racy the whores are us.
P. J. ORourke, American journalist, (1947–)

The quantum mechanical ground state energy of a particle of mass m in a
box (rectangular parallelepiped) with sides a, b, c, is given ([Wie73]) by

E =
h̄2 π2

2 m

(
1
a2 +

1
b2 +

1
c2

)
,

where h̄ = h/(2π), h being Planck’s constant.
What shape must the box have to minimize the energy E, subject to the

constraint that the volume V = a b c of the box is constant? What is the
minimum energy? Solve this problem by (a) the “direct” method, (b) the
method of Lagrange multipliers. Here is Ms. I. M. Curious’s solution.

(a) I. M. enters the energy expression E, setting A ≡ (h̄2 π2)/(2 m).
> restart: E:=A*(1/aˆ2+1/bˆ2+1/cˆ2);

E := A (
1
a2 +

1
b2 +

1
c2 )

The volume V = a b c is constant, so I. M. eliminates the variable c by setting
c=V/(ab). The new form of E is then displayed.

> c:=V/(a*b): E:=E;

E := A (
1
a2 +

1
b2 +

a2 b2

V 2 )

To determine the values of a and b which will extremize E, she sets ∂E/∂a=0
and ∂E/∂b=0 in eq1 and eq2 , respectively.

> eq1:=diff(E,a)=0; eq2:=diff(E,b)=0;

eq1 := A (− 2
a3 +

2 a b2

V 2 ) = 0 eq2 := A (− 2
b3 +

2 a2 b

V 2 ) = 0

For the extremum to be a minimum, I. M. notes that one must have [Ste87]

∂2E

∂a2 > 0, and
(

∂2E

∂a2

)(
∂2E

∂b2

)
−
(

∂2E

∂a∂b

)2

> 0.

The left-hand sides of these two conditions are now entered in eq3a and eq3b.
> eq3a:=diff(E,a,a);

eq3b:=diff(E,a,a)*diff(E,b,b)-diff(E,a,b)ˆ 2;

eq3a := A (
6
a4 +

2 b2

V 2 ) eq3b := A2 (
6
a4 +

2 b2

V 2 ) (
6
b4 +

2 a2

V 2 ) − 16 A2 a2 b2

V 4

Clearly ∂2E/∂a2 is positive, but it is not yet clear whether the second expression
in the above output is also positive. I. M. now solves eq1 and eq2 for a and b.

> sol:=solve({eq1,eq2},{a,b});

sol := {a = RootOf(−V + Z 3), b = RootOf(−V + Z 3)}, ....

Four possible solutions for a and b are generated in sol , all expressed as RootOf.
These four solutions are now converted to radical forms in sol2 .

> sol2:=seq(convert(sol[i],radical),i=1..4);

sol2 := {a = V (1/3), b = V (1/3)}, {a = −(−V )(1/3), b = (−V )(1/3)},

{a = −V (1/3), b = V (1/3)}, {a = (−V )(1/3), b = (−V )(1/3)}
Clearly, the first solution in sol2 is the desired one, which I. M. now assigns.
The values of a, b, c, and the minimum energy, Emin, immediately follow.

> assign(sol2[1]): a:=a; b:=b; c:=c; Emin:=E;

a := V (1/3) b := V (1/3) c := V (1/3) Emin :=
3 A

V (2/3)

To minimize the energy, the box must be cubical. That the energy is a minimum
follows on displaying eq3a and eq3b, which are both positive.

> eq3a:=eq3a; eq3b:=eq3b;

eq3a :=
8 A

V (4/3) eq3b :=
48 A2

V (8/3)

(b) Now, I. M. uses the method of Lagrange multipliers. The energy expression
E is entered along with g ≡ abc − V. The subsidiary condition is g=0.

> restart: E:=A*(1/aˆ2+1/bˆ2+1/cˆ2); g:=a*b*c-V;

E := A (
1
a2 +

1
b2 +

1
c2 ) g := a b c − V

I. M. forms a functional operator f to differentiate E + λ g with respect to an
arbitrary variable v, and then set the result equal to zero.

> f:=v->diff(E+lambda*g,v)=0:

Then f is used, with v equal to a, b, and c in eq1 , eq2 , and eq3 , respectively.
The relation g=0 is given in eq4 .

> eq||1:=f(a); eq||2:=f(b); eq||3:=f(c); eq||4:=g=0;

eq1 := −2 A

a3 + λ b c = 0 eq2 := −2 A

b3 + λ a c = 0

eq3 := −2 A

c3 + λ a b = 0 eq4 := a b c − V = 0

The sequence of four equations is solved for a, b, c, and λ, and each of the
solutions converted from RootOf forms to radical notation.

> sol:=solve({seq(eq||i,i=1..4)},{a,b,c,lambda}):
> sol2:=seq(convert(sol[i],radical),i=1..4);

sol2 := {b = V (1/3), c = V (1/3), a = V (1/3), λ =
2 A

V (5/3) }, ....

The first solution in sol2 is assigned and a, b, c, and Emin displayed.

> assign(sol2[1]): a:=a; b:=b; c:=c; Emin:=E;

a := V (1/3) b := V (1/3) c := V (1/3) Emin :=
3 A

V (2/3)

As I. M. expected, the answer agrees with that obtained with the direct method.

7.2.2 Erehwon Hydro Line

The fundamental concept in social science is Power, in the same
sense in which Energy is the fundamental concept in physics.
Bertrand Russell, British philosopher, mathematician, (1872–1970)

Erehwon Hydro has suspended a power line, of length L = 1.5 km and lin-
ear mass density ε = 1000 kg/km, across a deep, wide, gorge. Relative to the
origin at the gorge bottom, the two towers from which the cable is suspended
are at (−a/2, b) and (a/2, b), where a = 1.25 km and b = 1 km. Assuming
that the equilibrium shape of the cable is such as to minimize the potential en-
ergy, determine the cable shape and plot it. Take the gravitational acceleration
g=9.8/1000 km/s2. What is the distance between the lowest point in the cable
and the gorge bottom?

If an arclength element ds=
√

1 + (y ′)2 dx of cable at a point x is a distance
y(x) above the gorge bottom, the potential energy of the whole cable is

V =
∫

ε ds g y=
∫ a/2

−a/2
ε g y

√
1 + (y ′)2 dx.

After loading the VariationalCalculus package,
> restart: with(VariationalCalculus):

the integrand F =ε g y
√

1 + (y ′)2 of the potential energy V is entered.
> F:=epsilon*g*y(x)*sqrt(1+diff(y(x),x)ˆ2);

F := ε g y(x)

√
1 + (

d

dx
y(x))2

The subsidiary condition is that the length of the cable must be L, i.e.,∫
ds =

∫ a/2

−a/2

√
1 + (y ′)2 dx = L.

The integrand G=
√

1 + (y ′)2 of this constraint relation is entered,
> G:=sqrt(1+diff(y(x),x)ˆ2);

G :=

√
1 + (

d

dx
y(x))2

and the combination FF =F +λG formed, where λ is the Lagrange multiplier.
> FF:=simplify(F+lambda*G);

FF :=

√
1 + (

d

dx
y(x))2 (ε g y(x) + λ)

The EulerLagrange command is applied to FF and simplified. Since FF
doesn’t explicitly depend on x, in addition to the lhs of the Euler–Lagrange
equation, a first integral is generated, the integration constant being K1.

> eq:=simplify(EulerLagrange(FF,x,y(x)));

eq :=

⎧⎪⎨
⎪⎩

ε g y(x) + λ√
1 + (

d

dx
y(x))2

= K1,

ε g + ε g (
d

dx
y(x))2 − (

d2

dx2 y(x)) ε g y(x) − (
d2

dx2 y(x)) λ

(1 + (
d

dx
y(x))2)(3/2)

⎫⎪⎬
⎪⎭

The select command is used to extract the first integral relation.
> eq2:=select(has,eq,K[1])[1];

eq2 :=
ε g y(x) + λ√
1 + (

d

dx
y(x))2

= K1

The ODE in eq2 is analytically solved for y(x), assuming that ε > 0, g > 0.
> sol:=dsolve(eq2,y(x)) assuming epsilon>0,g>0;

Two equivalent solutions (not shown here) are produced in sol , the rhs of the
second solution being chosen, simplified, and assigned the name Y .

> Y:=simplify(rhs(sol[2]));

Y :=
1
2

(K1
2 + e

(− 2 ε g (x− C1)
K1

) − 2 e
(− ε g (x− C1)

K1
)
λ) e

( ε g (x− C1)
K1

)

ε g
The solution Y is converted to a trig form, and the combine command applied.

> Y2:=combine(convert(Y,trig));

Y2 :=
1
2
(K1

2 cosh(
ε g (x − C1 )

K1
) + K1

2 sinh(
ε g (x − C1 )

K1
)

+ cosh(
ε g (x − C1 )

K1
) − sinh(

ε g (x − C1 )
K1

) − 2 λ)
/
(ε g)

The cosh and sinh terms are collected in Y2 .
> Y3:=collect(Y2,[cosh,sinh]);

Y3 :=
1
2

(1+K1
2) cosh(

ε g (x − C1 )
K1

)

ε g
+

1
2

(K1
2 − 1) sinh(

ε g (x − C1 )
K1

)

ε g
− λ

ε g
To evaluate the constants K1, C1, and λ, three conditions are needed. Since
the cable will hang symmetrically, the slope at the center of the cable (x = 0)
must be zero. This is entered as boundary condition bc1 . At the end point
x=a/2, the vertical coordinate is b. This is entered in bc2 .

> bc1:=eval(diff(Y3,x),x=0)=0: bc2:=eval(Y3,x=a/2)=b:

The two boundary conditions are solved for λ and C1.
> sol2:=solve({bc1,bc2},{lambda,_C1}): assign(sol2):

After assigning sol2 , the cable shape is given by Y4 .
> Y4:=expand(Y3);

Y4 :=
cosh(

ε g x

K1
) K1

ε g
−

cosh(
1
2

a ε g

K1
) K1

ε g
+ b

The constant K1 remains to be evaluated. The constraint
∫ a/2

−a/2 Gdx = L is
applied in bc3 . To perform the integration, it is assumed that a > 0, ε > 0,
g > 0, and K1 > 0.

> bc3:=simplify(int(eval(G,y(x)=Y4),x=-a/2..a/2))=L

assuming a>0,epsilon>0,g>0,K[1]>0;

bc3 :=
e
(−1/2 a ε g

K1
)
K1 (−1 + e

(a ε g
K1

))
ε g

= L

bc3 is a transcendental equation, which must be solved numerically for K1. The
given parameter values are entered.

> epsilon:=1000: L:=1.5: a:=1.25: b:=1: g:=9.8/1000:

Then bc3 is numerically solved for K1, the value being labeled K1 . Evaluating
Y4 at K1 =K1 , the equilibrium shape of the cable is given by Y5 .

> K1:=fsolve(bc3,K[1]=0..10); Y5:=eval(Y4,K[1]=K1);

K1 := 5.751883654
Y5 := 0.5869269033 cosh(1.703789678 x) + 0.0476433492

The equilibrium shape Y5 , called a catenary, is plotted with constrained scal-
ing, the resulting picture being shown in Figure 7.4.

> plot(Y5,x=-a/2..a/2,thickness=2,view=[-a/2..a/2,0..b],

scaling=constrained,tickmarks=[3,3],labels=["x","y"]);

Figure 7.4: Equilibrium shape of the power line.

Evaluating Y5 at x=0, the lowest point on the cable
> height:=eval(Y5,x=0);

height := 0.6345702525
is about 0.635 km, or 635 m, above the gorge bottom.

7.3 Lagrange’s Equations

In classical mechanics, the Lagrangian L is defined by L = T − V , where T
is the kinetic energy and V the potential energy of the system of interest. If
L=L(qi, q̇i, t) where qi are the generalized coordinates (any set of coordinates
completely specifying the state of the system), q̇i the (generalized) velocity com-
ponents, and t the time, then Hamilton’s principle states that of all possible
motions the actual motion of the system over a time interval t0 to t1 is the one
for which

∫ t1
t0

L(qi, q̇i, t) dt is an extremum. Setting the variation of the integral
equal to zero leads to Lagrange’s equations of motion

∂L

∂qi
− d

dt

(
∂L

∂q̇i

)
= 0, (7.8)

which are equivalent to Newton’s law of motion. Since (7.8) is of the same
structure as Eq. (7.2), we can also make use of the EulerLagrange command.

7.3.1 Daniel’s Chaotic Pendulum

Out of chaos God made a world,
and out of high passions comes a people.
Lord Byron, English poet, (1788–1824)

Daniel’s engineering father has given him a toy for his birthday which basi-
cally consists of a simple pendulum (a small mass m attached to the end of a
light rod of length b), whose pivot point is attached to the rim of a wheel of
radius a which rotates at a constant angular velocity ω. The pendulum is in
the same vertical plane as the wheel and is free to rotate completely about its
pivot point. All frictional effects are neglected. This recipe will determine the
equation of motion for m, numerically solve the resulting nonlinear ODE, and
animate the motion. The plots and VariationalCalculus packages are loaded.

> restart: with(plots): with(VariationalCalculus):

If θ(t) is the angle that the pendulum rod makes with the vertical, the horizontal
coordinate of m at time t is x=a cos(ω t) + b sin(θ(t)) which is entered.

> x:=a*cos(omega*t)+b*sin(theta(t));

x := a cos(ω t) + b sin(θ(t))
The vertical coordinate of m at time t is y=a sin(ω t) − b cos(θ(t)).

> y:=a*sin(omega*t)-b*cos(theta(t));

y := a sin(ω t) − b cos(θ(t))
The horizontal (ẋ) and vertical (ẏ) components of m’s velocity are calculated.

> xdot:=diff(x,t); ydot:=diff(y,t);

xdot := −a sin(ω t) ω + b cos(θ(t)) (
d

dt
θ(t))

ydot := a cos(ω t) ω + b sin(θ(t)) (
d

dt
θ(t))

The kinetic (T = 1
2 m (ẋ2+ẏ2)) and potential (V =m g y) energy are determined.

> T:=simplify(m*(xdotˆ2+ydotˆ2)/2); V:=m*g*y;

T :=
1
2
m(a2 ω2 − 2 a sin(ω t) ω b cos(θ(t)) (

d

dt
θ(t))

+ 2 a cos(ω t) ω b sin(θ(t)) (
d

dt
θ(t)) + b2 (

d

dt
θ(t))2)

V := m g (a sin(ω t) − b cos(θ(t)))
The Lagrangian L=T−V is formed, T being simplified by applying the combine
command with the trig option.

> L:=combine(T,trig)-V;

L :=
m a2 ω2

2
− m a ω b (

d

dt
θ(t)) sin(ω t − θ(t)) +

1
2

m b2 (
d

dt
θ(t))2

− m g (a sin(ω t) − b cos(θ(t)))

The governing equation of motion, ode, is obtained by applying the
EulerLagrange command as follows. Because L depends explicitly on t, a
first integral is not generated with the EulerLagrange command.

> ode:=simplify(-EulerLagrange(L,t,theta(t))[1]/(m*b))=0;

ode := g sin(θ(t)) − a ω2 cos(ω t − θ(t)) + b (
d2

dt2
θ(t)) = 0

Since ode is a nonlinear ODE, it must be solved numerically. We will consider
a giant version of Daniel’s toy taking a=1 m and b=3 m. The wheel is rotated
at ω=2 rads/s and g=9.8 m/s2. The ODE is then displayed in ode2 .

> a:=1: b:=3: omega:=2: g:=9.8: ode2:=ode;

ode2 := 9.8 sin(θ(t)) − 4 cos(2 t − θ(t)) + 3 (
d2

dt2
θ(t)) = 0

For the initial condition, let’s take θ(0)=π/6 rads and θ̇(0)=0.
> ic:=theta(0)=Pi/6,D(theta)(0)=0:

Then, ode2 is numerically solved in sol, subject to the initial condition, for
θ(t). The output is given as a listprocedure.

> sol:=dsolve({ode2,ic},theta(t),type=numeric,
output=listprocedure):

Making use of sol , the next three command lines allow us to evaluate the
horizontal and vertical coordinates of m at an arbitrary time t.

> Theta:=eval(theta(t),sol):

> X:=t->eval(x,theta(t)=Theta(t)):

> Y:=t->eval(y,theta(t)=Theta(t)):

The circle command, found in the plottools library package, is used to plot a
thick red circle of radius a, centered on the origin, representing the wheel’s rim.

> c:=plottools[circle]([0,0],a,color=red,style=line,

thickness=2):
The line command (also in the plottools package) is used to create an arrow
operator l to draw the pendulum rod as a thick green line at arbitrary time t.

> l:=t->plottools[line]([a*cos(omega*t),a*sin(omega*t)],

[X(t),Y(t)],color=green,thickness=2):
The pointplot command is used to form an operator p to draw size 16 blue
circles at the pivot point and at m at arbitrary time t.

> p:=t->pointplot({[a*cos(omega*t),a*sin(omega*t)],
[X(t),Y(t)]},symbol=circle,symbolsize=16,color=blue):

The three graphs are superimposed in gr||i at a time t=0.1 i, and a do loop
used to generate plots for i from 0 to 200, i.e, for t=0 to 20 s.

> for i from 0 to 200 do

> t:=0.1*i:

> gr||i:=display({c,p(t),l(t)},labels=["x","y"]):
> end do:

The motion of the pendulum is animated by displaying the sequence of graphs,
using the insequence=true option.

> display(seq(gr||i,i=0..200),insequence=true);

A typical (at t=0.1 s) frame of the animation is shown in Figure 7.5. Click on
the computer plot and on the start arrow to see the entire pendulum motion.

Figure 7.5: Typical frame of the animation of Daniel’s chaotic pendulum.

For the choice of parameter values, the motion is quite irregular or “chaotic”
in appearance. You should try other values and see what happens.

7.3.2 Van Allen Belts

He could not see a belt without hitting below it.
Margot Asquith, British socialite about a former prime minister, (1864–1945)

As an interesting and non-trivial example, in this recipe we will determine
and animate the motion of a proton moving in the magnetic dipole field of the
earth. The animated motion is characteristic of the behavior of charged parti-
cles trapped in the Van Allen radiation belts surrounding the earth. [Bra68]

This recipe uses the plots, VariationalCalculus, and VectorCalculus pack-
ages which will generate several warnings unless suppressed, e.g., by using the
interface command to set the warnlevel to 0.

> restart: interface(warnlevel=0):

> with(plots): with(VariationalCalculus): with(VectorCalculus):

The magnetic field of the earth will be treated as a pure magnetic dipole oriented
along the z-axis and spherical polar coordinates (r, θ, φ) used. The velocity
vector �v of the proton at arbitrary time t is entered.

> v:=VectorField(<diff(r(t),t),r(t)*diff(theta(t),t),r(t)*

sin(theta(t))*diff(phi(t),t)>,’spherical’[r,theta,phi]);

v := (
d

dt
r(t)) er + r(t) (

d

dt
θ(t)) eθ + r(t) sin(θ(t)) (

d

dt
φ(t)) eφ

From Griffiths [Gri99], the spherical polar components of the vector poten-
tial �A at time t for a pure magnetic dipole are Ar = 0, Aθ = 0, and Aφ =
(µ0/4π) m sin(θ(t))/r2(t), where µ0 is the permeability of free space and m the
magnetic dipole moment (of the earth). The vector field �A is now entered.

> A:=VectorField(<0,0,(mu[0]/(4*Pi))*m*sin(theta(t))/r(t)ˆ2>,

’spherical’[r,theta,phi]);

A :=
1
4

µ0 m sin(θ(t))
π r(t)2

eφ

As a check on the form of �A, let’s calculate the magnetic field �B = ∇ × �A.
Removing the time dependence from �A and taking the curl yields the standard
expression [Gri99] for the magnetic dipole field.

> B:=Curl(subs({theta(t)=theta,r(t)=r},A));

B :=
1
2

µ0 m cos(θ)
r3 π

er +
1
4

sin(θ) µ0 m

r3 π
eθ

From Goldstein [GPS02], the Lagrangian for a particle of mass M and charge
q moving in the magnetic dipole field is given by L=(1/2) M (�v · �v) + q ( �A · �v).
This Lagrangian is now entered.

> L:=(1/2)*M*(v . v)+q*(A . v);

L :=
1
2

M ((
d

dt
r(t))2 + r(t)2 (

d

dt
θ(t))2 + r(t)2 sin(θ(t))2 (

d

dt
φ(t))2)

+
1
4

q µ0 m sin(θ(t))2 (
d

dt
φ(t))

π r(t)
Applying the EulerLagrange command to L, and specifying the time-dependent
spherical polar coordinates, yields three equations of motion for r, θ, and φ, as
well as two constants of the motion. The lengthy output is suppressed here.

> EL:=EulerLagrange(L,t,[r(t),theta(t),phi(t)]);

What do the constants of the motion tell us? First, let’s select the expression
in EL containing the constant K1.

> eq1:=select(has,EL,K[1])[1];

eq1 := M r(t)2 sin(θ(t))2 (
d

dt
φ(t)) +

1
4

q µ0 m sin(θ(t))2

π r(t)
= K1

The result eq1 can be used to determine the range of motion of the proton. For
given initial conditions, there are “forbidden” regions of space which the particle
cannot reach. The particle is “trapped” in an allowed region. This topic will
not be pursued here. The interested reader is referred to Bradbury. [Bra68]

Now, the expression in EL containing the constant K2 is selected, simplified,
and simplified further in eq2b with the substitution cos2(θ(t))=1 − sin2(θ(t)).

> eq2:=simplify(select(has,EL,K[2])[1]);

> eq2b:=algsubs(cos(theta(t))ˆ2=1-sin(theta(t))ˆ2,eq2);

eq2b := −1
2

M ((
d

dt
r(t))2 + r(t)2 (

d

dt
θ(t))2 + r(t)2 sin(θ(t))2 (

d

dt
φ(t))2) = K2

eq2b tells us that the speed of the particle remains constant with time. This
is hardly surprising, as it is well-known that a magnetic field can do no work
on a charged particle. Now, the motion of the proton in the magnetic dipole
field must be determined by solving the equations of motion numerically. We
remove the expressions involving the constants K1 and K2 from EL,

> eq3:=remove(has,EL,[K[1],K[2]]):

and write out the relevant equations in eq4 , eq5 , eq6 . Only eq4 is shown here.
> eq4:=expand(eq3[1]/M)=0; eq5:=expand(eq3[2]/M)=0;

eq6:=expand(eq3[3]/M)=0;

eq4 := −2 r(t) sin(θ(t))2 (
d

dt
φ(t)) (

d

dt
r(t))

− 2 r(t)2 sin(θ(t)) (
d

dt
φ(t)) cos(θ(t)) (

d

dt
θ(t)) − r(t)2 sin(θ(t))2 (

d2

dt2
φ(t))

− 1
2

q µ0 m sin(θ(t)) cos(θ(t)) (
d

dt
θ(t))

M π r(t)
+

1
4

q µ0 m sin(θ(t))2 (
d

dt
r(t))

M π r(t)2
= 0

To solve the three coupled nonlinear ODEs in eq4 , eq5 , and eq6 , we enter the
numerical values of the proton rest mass (M0 in kilograms), the proton charge
(q in Coulombs), the earth’s magnetic dipole moment (m in Joules/Tesla), the
permeability (µ0) of free space, the mean radius of the earth (RE in meters),
and the vacuum speed of light (c in meters/second).

> M0:=1.67*10ˆ(-27); q:=1.60*10ˆ(-19); m:=7.94*10ˆ(22);

mu[0]:=4*Pi*10ˆ (-7); RE:=6.37*10ˆ 6; c:=3*10ˆ 8;

M0 := 0.1670000000 10−26 q := 0.1600000000 10−18 m := 0.7940000000 1023

µ0 :=
π

2500000
RE := 0.637000000 107 c := 300000000

At time t=0, let’s take r(0)=2RE (radial distance twice the earth’s radius),
θ(0)=π/2 rads, φ(0)=0, ṙ(0)=0, φ̇(0)=0, and θ̇(0)=19 rads/sec.

> ic:=r(0)=2*RE,theta(0)=evalf(Pi/2),phi(0)=0,

D(r)(0)=0,D(phi)(0)=0,D(theta)(0)=19;

ic := r(0) = 0.1274000000 108, θ(0) = 1.570796327, φ(0) = 0,

D(r)(0) = 0, D(φ)(0) = 0, D(θ)(0) = 19
The mass of the proton is given by M = M0/

√
1 − β2, with β = v/c. In the

next two command lines, β and M are calculated.
> beta:=evalf(subs(ic,sqrt(D(r)(0)ˆ2+(r(0)*D(theta)(0))ˆ2

+(r(0)*sin(theta(0))*D(phi)(0))ˆ 2)/c));

β := 0.8068666667
> M:=M0/sqrt(1-betaˆ2);

M := 0.2826993436 10−26

In this simulation, the proton is traveling at about 8/10 the (vacuum) speed
of light, and its mass is about 5/3 times its rest mass. eq4 , eq5 , and eq6 are
expressed with the numerical values substituted, their outputs suppressed here.

> eq4:=evalf(eq4); eq5:=evalf(eq5); eq6:=evalf(eq6);

The three ODES are numerically solved in sol subject to the initial condition,
> sol:=dsolve({eq4,eq5,eq6,ic},{r(t),theta(t),phi(t)},

type=numeric,output=listprocedure):
and r(t), θ(t), φ(t) evaluated at arbitrary time t in R, Theta, and Phi.

> R:=eval(r(t),sol): Theta:=eval(theta(t),sol):

Phi:=eval(phi(t),sol):
For plotting purposes, three functional operators are now introduced to convert
from spherical polar coordinates to Cartesian coordinates at arbitrary time t.

> x:=t->R(t)*sin(Theta(t))*cos(Phi(t)):

> y:=t->R(t)*sin(Theta(t))*sin(Phi(t)):

> z:=t->R(t)*cos(Theta(t)):

The total time (tt) for the animation is taken to be 3.5 seconds, N =100 frames
will be used, and the time step will be tt/N .

> tt:=3.5: N:=100: step:=tt/N:

In gr1, the spacecurve command is used to plot the proton’s trajectory from
t=0 to t= tt. The coordinates x(t), y(t), and z(t) are normalized by dividing
by the radius RE of the earth. The trajectory is colored with the zhue option.

> gr1:=spacecurve([x(t)/RE,y(t)/RE,z(t)/RE],t=0..tt,

numpoints=2000,shading=zhue):
In gr2, the plot3d command is used in spherical coordinates to plot a bluish
colored sphere of radius 1 to represent the earth.

> gr2:=plot3d(1,theta=0..Pi,phi=0..2*Pi,coords=spherical,

color=COLOR(RGB,0.1,0.5,0.8),style=patchnogrid):
In the following do loop, the proton’s position is plotted at each time step and
superimposed on the graphs gr1 and gr2.

> for n from 0 to N do

> t:=step*n;

> gr3||n:=pointplot3d([x(t)/RE,y(t)/RE,z(t)/RE],style=point,

symbol=circle,symbolsize=16,color=red);
> pl||n:=display({gr1,gr2,gr3||n});
> end do:

Using the display command with the insequence=true option allows the pro-
ton’s motion to be animated.

> display(seq(pl||n,n=0..N),insequence=true,

scaling=constrained,axes=framed,labels=["x","y","z"],
orientation=[45,55],tickmarks=[3,3,3]);

On executing the last command line and clicking on the computer plot and on

the start arrow, the motion of the proton may be viewed. The opening frame
of the animation is shown in Figure 7.6, the proton represented by the small
circle at the bottom left of the figure.

Figure 7.6: Path traced out by proton in the magnetic dipole field of the earth.

In the animation, the proton spirals around the magnetic dipole field lines, un-
dergoing repeated reflections in the vicinity of the poles, and precessing around
the earth. The aurora occur in the vicinity of the turning points near the poles.

7.4 Rayleigh–Ritz Method

In Chapter 1, we noted that the simple harmonic oscillator, Bessel, Legendre,
and some other commonly occuring ODEs of physical interest, are all special
cases of the Sturm–Liouville (S-L) equation,

d

dx

[
p(x)

dy

dx

]
− q(x) y = −λw(x) y. (7.9)

for particular choices of the real functions p(x), q(x), and w(x). w(x) is taken
to be non-negative over the range x = a to b of interest. In boundary-value
problems where y satisfies the Sturm–Liouville boundary conditions (y or its
derivative vanish at a and b), y is the eigenfunction and λ the real eigenvalue.

The S-L equation can be formulated as a variational problem. Suppose that
we want to extremize the integral

I[y] =
∫ b

a

[p (y ′)2 + q y2] dx, subject to J [y] =
∫ b

a

w y2 dx = 1. (7.10)

The form that y must take is determined by solving the Euler–Lagrange (E-L)
equation with F = p (y ′)2 + q y2 − λw y2, where λ is the Lagrange multiplier.
Substituting F into the E-L equation just yields the S-L equation, (7.9). So the
function y which makes I[y] an extremum subject to the subsidiary condition
on J [y] is an eigenfunction of the S-L equation. The Lagrange multiplier λ is

the corresponding eigenvalue, while the subsidiary condition corresponds to the
normalization condition on the eigenfunctions.

Now form the quantity Λ[y]=I[y]/J [y]. Extremizing Λ is exactly equivalent
to extremizing I with the subsidiary condition on J . For S-L boundary condi-
tions, it can be shown that the value of Λ for y = yn, an eigenfunction of the
S-L equation, is equal to the eigenvalue λn, i.e., Λ[yn] =λn. This latter result
is the basis of the Rayleigh–Ritz method for estimating eigenvalues.

For example, for atomic or molecular systems more complicated than the
hydrogen atom, an exact analytical determination of the eigenvalues (energy
levels) and eigenfunctions is not possible, so one must resort to approximate
techniques. The Rayleigh–Ritz method is one such approach.

In this method one introduces a “trial function” φ, satisfying the boundary
conditions. For the lowest eigenvalue λ1 (ground state energy for molecular
systems), it can be shown that Λ[φ] ≥ λ1, i.e., Λ provides an upper bound to
the lowest eigenvalue. By introducing one or more adjustable parameters into
φ, the estimate of λ1 can usually be improved, the estimate converging on λ1
from above. The Rayleigh–Ritz method can also be used to estimate higher
eigenvalues, but the estimate does not necessarily converge to the exact answer
from above. In the recipes presented here, we will only estimate the lowest
eigenvalue. The text Mathematics in Physics and Engineering by Irving and
Mullineux [IM69] deals with estimating higher eigenvalues.

7.4.1 I. M. Estimates a Bessel Zero

To arrive at a just estimate of a renowned man’s character
one must judge it by the standards of his time, not ours.
Mark Twain, American author, (1835–1910)

This recipe, provided by Ms. I.M. Curious, solves the following problem, which
has often appeared in various guises on my mathematical physics exams.

The Bessel function of order 1 satisfies the equation
d2J1

dr2 +
1
r

dJ1

dr
+ (k2 − 1

r2 )J1 = 0.

Given that J1(r=0)=0, and noting that if we impose the boundary condition
J1(r=1)=0, then k must be a zero of J1(kr), use the Rayleigh–Ritz method with
no adjustable parameters to obtain an approximate value of the first zero of J1.
What is the percentage error in your answer when compared to the exact value?
Improve your estimate by including one adjustable parameter. Generally, the
agreement between the trial function and the exact solution is not nearly as
good as between the eigenvalue estimate and the exact eigenvalue. Confirm
that this is the case here by plotting the trial function with one adjustable
parameter and the exact first-order Bessel function solution.

I. M. begins by entering the general form of Λ[φ(r)]. The range of interest
here is from r=0 to 1, so the integrals are taken to have these limits. From the
structure of Λ, she notes that the trial function need not be normalized as the
normalization constant will obviously cancel out.

> restart:

> Lambda:=int(p(r)*diff(phi(r),r)ˆ2+q(r)*phi(r)ˆ2,r=0..1)

/int(w(r)*phi(r)ˆ 2,r=0..1);

Λ :=

∫ 1

0
p(r) (

d

dr
φ(r))2 + q(r) φ(r)2 dr∫ 1

0
w(r) φ(r)2 dr

The first-order Bessel equation can be put into Sturm–Liouville form by choos-
ing p(r)=r, q(r)=1/r, w(r)=r, and noting that λ=k2.

> p(r):=r: q(r):=1/r: w(r):=r:

I. M. chooses φ = r(1 − r)(1 + c r) as her2 trial function, with c an adjustable
parameter which she will set to 0 in the first part of the problem. The boundary
conditions are clearly satisfied, i.e., φ=0 at r=0 and 1.

> phi(r):=r*(1-r)*(1+c*r):

The trial function φ is automatically substituted into Λ, which on being sim-
plified takes the following form.

> Lambda:=simplify(Lambda);

Λ :=
14 (7 c2 + 18 c + 15)

5 c2 + 16 c + 14
To answer the first part of the question, I. M. evaluates Λ with c=0.

> Lambda0:=eval(Lambda,c=0);

Λ0 := 15
Then the estimated value of k is obtained by taking the square root of Λ0.

> k0:=evalf(sqrt(Lambda0));

k0 := 3.872983346
The exact (numerical) k value is obtained by using the BesselJZeros command.

> kexact:=BesselJZeros(1.0,1);

kexact := 3.831705970
I. M. now calculates the percentage error, 100 (k0 − kexact)/kexact , and finds
it to be about 1.08%.

> percenterror0:=((k0-kexact)/kexact)*100;

percenterror0 := 1.077258441
Now, she adjusts the parameter c to minimize the value of Λ. This is accom-
plished by calculating dΛ/dc and setting the result equal to 0.

> eq:=diff(Lambda,c)=0;

eq :=
14 (14 c + 18)

5 c2 + 16 c + 14
− 14 (7 c2 + 18 c + 15) (10 c + 16)

(5 c2 + 16 c + 14)2
= 0

2Since the form of φ is deliberately not specified in the wording of the problem, I do get
a variety of possible forms submitted by my students, sometimes even trial functions that do
not satisfy the boundary conditions!

Then, eq is solved for c, and put into decimal form by applying the floating
point evaluation command.

> sol:=evalf(solve(eq));

sol := −0.3055081542, −1.785400936
Two values are generated for c. To decide on which value will minimize Λ, I. M.
plots Λ over the range c=−2 to 1. The resulting picture is shown in Figure 7.7.

> plot(Lambda,c=-2..1,labels=["c","Lambda"]);

20

30

40

50

Lambda

–2 –1 0 c 1

Figure 7.7: Λ versus c.

From the figure, the value c � −1.79 must be ruled out as then Λ � 50 > Λ0.
Λ should be evaluated with c � −0.31. This is done in Λ1, the number being
slightly lower than Λ0=15.

> Lambda1:=eval(Lambda,c=sol[1]);

Λ1 := 14.84137598
Then, k1 =

√
Λ1 is calculated and the percentage error determined once again.

> k1=sqrt(Lambda1);

k1 := 3.852450646
> percenterror:=((k1-kexact)/kexact)*100;

percenterror := 0.5413952992
With one adjustable parameter included, the estimated value of k, i.e., k1 , now
only differs by 0.54% from the exact value. The corresponding trial function is
given by φ1.

> phi1:=eval(phi(r),c=sol[1]);

φ1 := r (1 − r) (1 − 0.3055081542 r)
To finish the problem, I. M. plots φ1 and the exact Bessel solution in Fig. 7.8.

> plot([phi1,BesselJ(1,kexact*r)],r=0..1,color=[red,blue],

labels=["r","phi"]);

Figure 7.8: Comparison of trial function (lower curve) and exact solution (top).

She observes that although the estimated value of k is extremely good, the trial
function φ1 doesn’t do a very good job of representing the exact first-order
Bessel function solution over the range r=0 to 1. Part of the discrepancy could
be due to the fact that φ1 and J1 are not similarly normalized.

7.4.2 I. M. Estimates the Ground State Energy

The only difference between a tax man and a taxidermist
is that the taxidermist leaves the skin.
Mark Twain, American author, (1835–1910)

In the last example, the trial function contained one adjustable parameter.
In the following problem, for which I. M. Curious will provide her recipe as a
solution, there are two parameters which must be adjusted to minimize Λ.

Use the Rayleigh–Ritz procedure and the trial function φ = e−x2
to esti-

mate the ground state energy E1 of a particle satisfying the one-dimensional
Schrödinger equation (in appropriate units)

d2ψ

dx2 − x4 ψ = −E ψ, −∞ < x < ∞.

Improve your estimate by considering φ = e−x2
(1 + c1 x2 + c2 x4), with two

adjustable parameters c1 and c2 .
I. M. enters the general form of Λ, the range now from x=−∞ to ∞.
> restart: with(plots):

> Lambda:=int(p(x)*diff(phi(x),x)ˆ2+q(x)*phi(x)ˆ2,x=-infinity

..infinity)/int(w(x)*phi(x)ˆ 2,x=-infinity..infinity);

Λ :=

∫ ∞

−∞
p(x) (

d

dx
φ(x))2 + q(x) φ(x)2 dx∫ ∞

−∞
w(x) φ(x)2 dx

From the given ODE, she identifies p(x)=1, q(x)=x4, and w(x)=1, which are
now entered. The eigenvalue λ=E.

> p(x):=1: q(x):=xˆ4: w(x):=1:

I. M. enters the trial wave function with the two adjustable parameters. The
first part of the question can be answered by setting c1 =0, c2 =0.

> phi(x):=exp(-xˆ2)*(1+c1*xˆ2+c2*xˆ4);

φ(x) := e(−x2) (1 + c1 x2 + c2 x4)
φ(x) is automatically substituted into Λ, which on simplifying is as follows.

> Lambda:=simplify(value(Lambda));

Λ :=
13995 c2 2 + 10248 c2 c1 − 1248 c2 + 3472 c1 2 + 4864 − 128 c1

16 (256 + 105 c2 2 + 120 c2 c1 + 128 c1 + 96 c2 + 48 c1 2)
The estimate of the ground state energy with no adjustable parameters is de-
termined in Λ0.

> Lambda0:=evalf(eval(Lambda,{c1=0,c2=0}));
Λ0 := 1.187500000

So, I. M. estimates the ground state energy E1 to be about 1.19 units. Now,
she considers the situation when the parameters are non-zero. In this case, she
must minimize Λ with respect to both parameters. This is done by imposing
the conditions ∂Λ/∂c1 =0 and ∂Λ/∂c2 =0, which is done in C1 and C2 .

> C1:=simplify(diff(Lambda,c1))=0;

C2:=simplify(diff(Lambda,c2))=0;

C1 := −2(−68352 c2 + 18855 c2 3 + 19200 c2 2 c1 − 24576 c2 c1

+ 20976 c2 2 + 2352 c2 c1 2 − 40960 c1 − 14080 c1 2 + 20480)

/(256 + 105 c2 2 + 120 c2 c1 + 128 c1 + 96 c2 + 48 c1 2)2 = 0

C2 := 6(64000 c2 + 6285 c2 2 c1 + 37600 c2 c1 + 15360 c2 2

+ 6400 c2 c1 2 + 19712 c1 + 9728 c1 2 + 784 c1 3 − 8192)

/(256 + 105 c2 2 + 120 c2 c1 + 128 c1 + 96 c2 + 48 c1 2)2 = 0
The formidable nonlinear algebraic equations C1 and C2 must be numerically
solved for c1 and c2 . To guide her in arriving at the correct solution, I. M. forms
a functional operator gr to apply the implicitplot command to a specified
algebraic equation A, the color of the curve being dictated by the choice of B.
To obtain smooth curves, the number of plotting points is increased to 2000.

> gr:=(A,B)->implicitplot(A,c1=-10..10,c2=-10..10,

numpoints=2000,color=B):
A plot of c2 versus c1 satisfying the equation C1 follows on taking A= C1 in
gr. The curve will be colored blue. Similarly, a green curve is generated for the
choice A= C2 . The two curves are superimposed with the display command,
the resulting picture being shown in Figure 7.9.

> display({gr(C1,blue),gr(C2,green)});

Figure 7.9: Possible values of c1 and c2 occur at intersection points.

Within the range of the plot, I. M. spots three intersection points. Applying
the floating point solve command to C1 and C2 without specifying the search
range for c1 and c2 ,

> s1:=fsolve({C1,C2},{c1,c2});
s1 := {c1 = −7.524875339, c2 = 4.879077669}

yields the intersection point in the second quadrant of Figure 7.9. To obtain
the other two intersection points, I. M. now specifies the appropriate plotting
ranges in the fsolve command in s2 and s3

> s2:=fsolve({C1,C2},{c1,c2},{c1=0..2,c2=-2..2});
s2 := {c1 = 0.5221933646, c2 = −0.05771753990}

> s3:=fsolve({C1,C2},{c1,c2},{c1=-5..0,c2=-2..2});
s3 := {c2 = −0.2086941568, c1 = −3.169995132}

An operator L is created to evaluate λ for a specified s, which I. M. then applies
to s1 , s2 , and s3 in Λ1, Λ2, and Λ3.

> L:=s->eval(Lambda,s):

> Lambda1:=L(s1); Lambda2:=L(s2); Lambda3:=L(s3);

Λ1 := 16.71552532 Λ2 := 1.061092905 Λ3 := 7.535881788
The estimates Λ1 and Λ3 are rejected as both are larger than Λ0 � 1.19. The
estimate Λ2 � 1.06 is lower than Λ0, so represents an improved estimate of
the ground state energy. It would appear from Iam’s recipe that her estimates
are converging on the value E1 =1. It is left as an exercise for you to confirm
whether this conclusion is correct or not.

7.5 Supplementary Recipes
07-S01: Geodesic
The curve of shortest length joining two points is called a geodesic. Show that
the geodesic on the surface of a right circular cylinder of radius a is a helix. Plot
the geodesic on a cylinder of radius a=2 between the points (z=1/2, θ=π/8)
and (5/2, π/2), where z is the cylinder axis coordinate and θ the polar angle.

07-S02: Laws of Geometrical Optics
Use Fermat’s principle to prove the following geometrical optics relations:

(a) A light ray incident at an angle i to the normal on a planar mirror is
reflected back into the same medium at an angle r ′ = i to the normal.

(b) Consider a light ray traveling from a medium with refractive index n1
through a planar interface into a medium with refractive index n2. If
the light ray is incident at an angle i, the angle of refraction r in the
second medium is given by Snell’s law: n1 sin i = n2 sin r. Both angles
are measured with respect to the normal to the interface.

07-S03: Bending of Starlight
According to the theory of general relativity, the trajectory of starlight travel-
ing in the spherically symmetric static field of the sun is such as to minimize
the integral

I =
∫ √

(dr/γ)2 + (r dθ)2/γ

where (r, θ) are polar coordinates and γ = 1−(2 Gms)/(c2 r), G being the grav-
itational constant, ms the mass of the sun, and c the vacuum speed of light.
Using the variational approach and setting u = 1/r, prove that the differential
equation of the trajectory can be written in the form

d2u

dθ2 + a u = b u2,

where a and b remain to be identified. Taking G=6.673×10−11 N·m2/kg2, ms=
1.99×1030 kg, c=2.997×108 m/s, and the sun’s radius Rs=6.96×108 m, numer-
ically solve the nonlinear ODE for u(θ), taking u(0)=1/(10Rs) and du(0)/dθ=
0. Then use the spacecurve command to plot (r(θ) cos(θ), r(θ) sin(θ)) for
θ = −π/2 to π/2. Include the sun in your figure.

07-S04: Another Refractive Index
Using Fermat’s principle, prove that light rays in a medium with refractive
index n(x, y)=1/y will follow a path which is the arc of a circle.

07-S05: Mirage
Assuming that the refractive index varies linearly with height in the following
way, n(x, y) = n0 (1 + α y) with n0 > 0 and α > 0, use Fermat’s principle to
determine the angle θ by which a point P is apparently lowered when viewed
from a point P ′ at the same height as P at a horizontal distance d. The
parameter α is sufficiently small that you may take α d << 1. This problem is
relevant to the phenomenon of mirages observed in hot desert regions.

07-S06: A Constrained Extremum
Determine a function y(x) for which I =

∫ π

0 ((y ′)2 − y2) dx is an extremum,
subject to N =

∫ π

0 y dx=1, and y(0)=0, y(π)=1. Plot y(x).

07-S07: Maximum Volume
Find the maximum value of the volume V = x y z, subject to the condition
x2/a2 + y2/b2 + z2/c2 =1, with a, b, and c positive. Use (a) the direct method,
(b) the Lagrange multiplier method.

07-S08: Eigenvalue Estimate
Taking a trial function of the form y=A0 + A1 x + A2 x2 + A3 x3, estimate the
smallest eigenvalue k of y ′′ + k2 y =0, subject to y(0)=y(1)=0. Compare the
estimate and the trial function with the exact results.

07-S09: Surface of Revolution
Consider a surface of revolution generated by revolving a curve y(x) about the
x-axis. The curve is required to pass through the fixed end points (x=0, y=1)
and (x = 1, y = 2). Determine what shape y(x) must have so that the area of
the resulting surface will be a minimum. Plot the curve.

07-S10: Dido Wasn’t a Dodo
Given a closed planar curve C of a fixed length L, what shape should the curve
have to maximize the enclosed area A? The answer, known as the isoperimetric
theorem, is a circle. Evidently, this theorem has been known from about 900
BC, and according to Virgil’s Aneid was applied by Queen Dido in a practical
way to establish the city of Carthage (now Tunisia) in North Africa. The
local ruler, King Jambas, agreed to sell her all the land that she could enclose
within a bull’s hide. She cleverly had the hide cut into thin strips which were
joined end to end to form a long “string” and made use of a variation on the
isoperimetric theorem. She cleverly maximized A by selecting a straight portion
of the Mediterranean coast and laying the string out in a semicircle with the
string’s two ends touching the coast.

Given that A= 1
2

∮
C

(x dy − y dx) for a closed planar curve C, make use of
the Euler–Lagrange equation to prove the isoperimetric theorem.

07-S11: Another Approach to the String Equation
In recipe 04-1-1, the wave equation for small transverse vibrations of a light
flexible horizontal string under tension and fixed on its ends was derived us-
ing Newton’s second law of motion. Another approach to deriving the wave
equation is to use the variational method.

It can be shown that a necessary condition for y=y(x, t) to be an extremal
of
∫ t2

t1

∫ x2

x1
F (x, t, y, yx, yt) dx dt is

∂F

∂y
− ∂

∂x

(
∂F

∂yx

)
− ∂

∂t

(
∂F

∂yt

)
= 0.

Here yx ≡ ∂y/∂x, yt ≡ ∂y/∂t, x1 and x2 are fixed end points, and t1 and t2 are
specified times. Making use of this extension of Lagrange’s equation, derive the

wave equation for small vibrations of a flexible string, fixed at its end points
x=0 and a, and under tension τ .

07-S12: Betsy Bug’s Ride
Betsy bug is perched on a bead which slides along a smooth wire bent in the
shape of a stylized “W”. I.e., the bead’s vertical coordinate at time t is z(t)=
−c1 r(t)2 + c2 r(t)4, with c1 > 0, c2 > 0, and r(t) the radial distance from the
vertical symmetry axis. The wire is rotating about it’s symmetry axis with
angular velocity ω radians/s, so the bead’s horizontal coordinates at time t are
x(t)=r(t) cos(ω t), y(t)=r(t) sin(ω t).

(a) Using the Lagrangian approach, find the equation of motion for r(t).

(b) Show that there is critical frequency ωcr =
√

2 g (2 c2 r(0)2 − c1), where
g is the acceleration due to gravity, for which the bead will not move
relative to the wire.

(c) Taking c1 =1 m−1, c2 =1/2 m−3, gravitational acceleration g=9.8 m/s2,
r(0)= 2 m, ṙ(0)=0 m/s, determine the critical frequency.

(d) Taking ω=0.1 ωcr, numerically solve the equation of motion for r(t).

(e) Animate the motion of the bead and the wire.



Part III

THE DESSERTS

The reasonable man adapts himself to the world;
the unreasonable one persists in trying to adapt the
world to himself. Therefore, all progress depends

on the unreasonable man.
George Bernard Shaw, Anglo-Irish playwright, Man and Superman, (1903)

Perfection is achieved not when you have nothing more
to add, but when you have nothing left to take away.

Antoine de Saint-Exupery, French pilot and author, (1900–1944)

I may not have gone where I intended to go,
but I think I have ended up where I intended to be.

Douglas Adams, English humorist and science fiction writer, (1952–2001)



Chapter 8

NLODEs & PDEs of Physics

So far, this gourmet selection of computer algebra recipes has emphasized phys-
ical examples governed by linear ODEs and PDEs. However, much of modern
scientific research involves phenomena described by nonlinear differential equa-
tions, i.e, equations which are not linear in the dependent variable(s). In the
Desserts, I will give you a small taste of this intellectually delectable area of
mathematical physics. If you enjoy the nonlinear recipes that follow, and crave
more, you should consult [EM00] for a much deeper treatment of the subject.

This chapter illustrates a few of the basic analytic (exact and approximate)
and graphical methods for solving nonlinear ODEs and PDEs (NLODEs and
NLPDEs). Basic numerical methods will be covered in Chapter 9. A compre-
hensive survey of analytic and numerical approaches may be found in Daniel
Zwillinger’s Handbook of Differential Equations [Zwi89].

8.1 Nonlinear ODEs: Exact Methods

In Chapter 1, we encountered a simple nonlinear ODE, describing the motion
of a falling badminton bird acted upon by a nonlinear drag force, which had an
exact analytic solution. Only a handful of NLODEs of physical interest can be
exactly solved using elementary techniques. Here are a few more examples of
increasing mathematical complexity.

8.1.1 Jacob Bernoulli and the Nonlinear Diode

I recognize the lion by his paw.
Jacob Bernoulli, Swiss mathematician, after reading an anonymous solution to
a problem that he realized was Newton’s solution, (1654–1705)

What does Jacob Bernoulli who lived some 300 years ago have to do with
a modern nonlinear diode? As you will shortly see, the governing circuit equa-
tion in the following recipe is the nonlinear Bernoulli equation which Jacob
discovered in another context in 1690 and solved in 1696. Bernoulli’s equation

is a first order nonlinear ODE of the form

y ′ + f1(x) y = f2(x) yn (8.1)

which can be cast into a linear ODE, and thus solved, by introducing a new
dependent variable p = 1/yn−1. Bernoulli’s equation is known to Maple and
thus can be easily solved without having to carry out the details of the variable
transformation. Now let’s formulate the nonlinear diode problem.

A linear capacitor C is connected in series to a nonlinear diode, a circuit
element which has a current (i)–voltage (v) relation of the form i=a v + b vn,
where a and b are positive constants and n=2, 3, 4, · · · Given that the voltage
across the capacitor at time t=0 is v(0)=V , what is the voltage v(t) for t > 0?
Express the solution in dimensionless form in terms of a single parameter β,
which remains to be identified. Plot the analytic result for n=2 to 5 and β=2.

To solve this problem, the PDEtools library package is loaded. This contains
the dchange command which will be used to cast the governing circuit equation
and the voltage into dimensionless form.

> restart: with(PDEtools):

From the definition of capacitance, the charge q on the capacitor at time t is
q = C v(t). So, the current in the circuit is i = −dq/dt = a v(t) + b v(t)n.
These two basic relations are entered, the expression for q being automatically
substituted into i.

> q:=C*v(t);

q := C v(t)
> i:=-diff(q,t)=a*v(t)+b*v(t)ˆn;

i := −C (
d

dt
v(t)) = a v(t) + b v(t)n

The above ODE is clearly of Bernoulli’s form with y ≡ v, f1 ≡ a/C and
f2 ≡ −b/C. It can be transformed into dimensionless form by introducing a
new time variable τ =a t/C and a dimensionless voltage variable y(τ)=v(t)/V .

> tr:={t=C*tau/a,v(t)=V*y(tau)};

tr := {t =
C τ

a
, v(t) = V y(τ)}

The above transformation is applied to i in the following dchange command.
> ode:=dchange(tr,i,[tau,y(tau)]);

ode := −a V (
d

dτ
y(τ)) = a V y(τ) + b (V y(τ))n

Then ode is divided by −a V and the result expanded.
> ode:=expand(ode/(-a*V));

ode :=
d

dτ
y(τ) = −y(τ) − b (V y(τ))n

a V
Introducing a dimensionless parameter β through the substitution b = a β/V (n−1)

and simplifying with the symbolic option produces the desired dimensionless
nonlinear ODE in ode2 .

> ode2:=simplify(subs(b=a*beta/Vˆ(n-1),ode),symbolic);

ode2 :=
d

dτ
y(τ) = −y(τ) − β y(τ)n

Note that this equation now only depends on one parameter, β, and that the
initial condition at τ = t=0 is y(τ)=1. Instead of mimicking a hand calculation,
and making the suggested dependent variable substitution, I will directly use
the dsolve command. First let’s enter the following infolevel command to
give us information on what approach Maple takes in solving ode2 .

> infolevel[dsolve]:=2:

Then, ode2 is solved for y(τ), subject to the initial condition y(0)=1.
> sol:=simplify(dsolve({ode2,y(0)=1},y(tau)));

Methods for first order ODEs:
— Trying classification methods —
trying a quadrature
trying 1st order linear
trying Bernoulli
<− Bernoulli successful

sol := y(τ) = (−β + e((n−1) τ) + e((n−1) τ) β)(−
1

n−1 )

The NLODE is identified as a Bernoulli equation and successfully solved. To
plot the solution, let’s enter the suggested value β=2.

> beta:=2:

The rhs of sol is now plotted for n= 2, 3, 4, 5, a different color being assigned
to each curve. The resulting graph is shown in Figure 8.1, the bottom curve
corresponding to n = 2, the next highest curve to n = 3, and so on. Because
y(τ) < 1 as τ increases, the curves decrease less rapidly as n is increased.

> plot([seq(rhs(sol),n=2..5)],tau=0..2,color=[red,green,blue,

black],thickness=2,labels=["tau","y"],tickmarks=[2,2])

y

Figure 8.1: Normalized voltage as a function of time.

8.1.2 The Chase

I wouldn’t think of asking you to lie;
you haven’t the necessary diplomatic training.
A line spoken by one of the characters in the movie Sea Chase (1955)

A classic problem [Dav62] in the history of nonlinear ODEs involves the curve
of pursuit. This is the trajectory generated by a point P which moves in such
a way that its direction of motion is always towards a second point P ′, con-
strained to move along a prescribed path. This type of problem originated with
Leonardo da Vinci in the 15th century, but the example in the following recipe
is due to the French hydrographer Pierre Bouguer who published his solution in
1732. Bouguer’s formulation is expressed in terms of one ship pursuing another,
the latter moving along a straight line.

I will phrase the problem somewhat differently. Referring to the lhs of
Figure 8.2, rascally Roger Rabbit is being pursued by ferocious Freddy Fox.

freddy fox

roger rabbit

capture?

freddy’s
  path

y

xa0

(x,y)

(a,Y)

a-x
Y-y

0

0.5

1

y

0.2 0.4 x 0.8 1

Figure 8.2: Left: Geometry for pursuit. Right: Freddy’s calculated path.

Roger, who is initially at the point x=a, y=0, runs at a constant speed along
the vertical line x = a. Freddy Fox, who is initially at the origin O, pursues
Roger by constantly aiming at Roger’s current position. Freddy’s constant
speed is n times that of Roger’s, with n > 1. The problem is to derive the
equation y(x) of Freddy’s path (indicated schematically in the picture) and
determine the theoretical point of capture. If a=1 km and n=3/2, will Roger
escape capture by diving into his hole which is located 1195 meters from Roger’s
starting point. Plot Freddy’s path up to the theoretical point of capture. So
let the chase begin!

The assumption that a > 0, x ≥ 0, x < a, and n > 1 is entered.
> restart: assume(a>0,x>=0,x<a,n>1):

From the figure, the slope of Freddy’s trajectory when he is at the point
(x, y(x)) and Roger is at (a, Y ) is given by dy/dx = (Y − y(x))/(a − x).



8.1. NONLINEAR ODES: EXACT METHODS 295

> slope:=diff(y(x),x)=(Y-y(x))/(a-x);

slope :=
d

dx
y(x) =

Y − y(x)
a − x

The slope equation is solved for Y ,

> Y:=solve(slope,Y);

Y := (
d

dx
y(x)) a − (

d

dx
y(x)) x + y(x)

and the x derivative of Y then taken, and factored.

> Yder:=factor(diff(Y,x));

Yder := (
d2

dx2 y(x)) (a − x)

The arclength along Freddy’s path is ds=
√

(dx)2 + (dy)2 and his speed, ds/dt,
is n times Roger’s speed, dY/dt. So, n (dY/dx)=

√
1 + (dy/dx)2 is entered.

> ode:=n*Yder=sqrt((diff(y(x),x))ˆ2+1);

ode := n (a − x) (
d2

dx2 y(x)) =

√
(

d

dx
y(x))2 + 1

Freddy’s path is described by a second order NLODE which, amazingly, can
be solved exactly. Although the dsolve command could be applied directly to
ode, the subsequent manipulations to put the solution in a compact form are
messy. Instead, let’s mimic the approach that we would undertake by hand.
Making the substitution dy(x)/dx = p(x) reduces ode to a first order ODE in
p(x) displayed in ode2 .

> ode2:=subs(diff(y(x),x)=p(x),ode);

ode2 := n (a − x) (
d

dx
p(x)) =

√
p(x)2 + 1

Freddy’s initial slope is zero, i.e., p(0)=0. ode2 is now analytically solved for
p(x), subject to the initial slope condition. The right-hand side of the solution
is then taken and the result equated to dy/dx. Occasionally, on executing the
worksheet, the form of the output will differ in ode3 from what is reproduced
here in the text. This will not affect the shape of Freddy’s trajectory or the
theoretical point of capture.

> ode3:=diff(y(x),x)=rhs(dsolve({ode2,p(0)=0},p(x)));

ode3 :=
d

dx
y(x) = −sinh(

ln(a − x) − ln(a)
n

)

ode3 is analytically solved for y(x), subject to the initial condition y(0) = 0.
Again, the right-hand side of the solution is taken. The equation of Freddy’s
path is then displayed in the output of the following command line.

> y:=rhs(dsolve({ode3,y(0)=0},y(x)));

y := −nx (a − x)(
1
n )

2 (n + 1) a( 1
n )

+
a n (a − x)(

1
n )

2 (n + 1) a( 1
n )

+
(− a n

2 (n − 1)
+

nx

2 (n − 1)
) a( 1

n )

(a − x)(
1
n )

− a n

2 (n + 1)
+

a n

2 (n − 1)
Next, let’s substitute x=a − Xn into y and expand the result.

> y:=expand(subs(x=a-Xˆn,y));

y :=
n (Xn)(

1
n ) Xn

2 (n + 1) a( 1
n )

− a( 1
n ) nXn

2 (Xn)(
1
n ) (n − 1)

− a n

2 (n + 1)
+

a n

2 (n − 1)

The previous substitution is reversed by setting X =(a − x)1/n in y. Applying
the combine command yields a compact equation describing Freddy’s trajectory.

> y:=combine(subs(X=(a-x)ˆ(1/n),y));

y :=
(a − x)(

1
n + 1) a(− 1

n ) n

2 (n + 1)
− (a − x)(1 − 1

n ) a( 1
n ) n

2 (n − 1)
− a n

2 (n + 1)
+

a n

2 (n − 1)
The theoretical point of capture, Yc, follows on evaluating y at x=a.

> Yc:=simplify(eval(y,x=a));

Yc :=
a n

n2 − 1
Roger would be captured at Yc = (a n)/(n2 − 1), unless he gets to his hole
first. So, does he avoid capture? Substituting a = 1 and n = 3/2 into Yc and
multiplying by 1000 to convert the result into meters,

> Yc:=evalf(subs({n=3/2,a=1},Yc))*1000;
Yc := 1200.000000

yields a value for Yc of 1200 meters. Fortunately, for Roger, his hole was only
1195 meters from his starting point, so he survives for another day. To plot
Freddy’s path up to the point of theoretical capture, the values n = 3/2 and
a=1 are substituted into y, yielding the result shown in y2 .

> y2:=subs({n=3/2,a=1},y);

y2 :=
3 (1 − x)(5/3)

10
− 3 (1 − x)(1/3)

2
+

6
5

Freddy’s trajectory y2 is then plotted up to the theoretical point of capture,

> plot(y2,x=0..1,thickness=2,numpoints=200,labels=["x","y"]);

the result being shown on the right-hand side of Figure 8.2. Of course, if Roger’s
hole were not conveniently close, he would have had to take evasive action, lead-
ing to a much more difficult pursuit problem to solve.

8.1.3 Not As Hard As It Seems

Often, the less there is to justify a traditional custom,
the harder it is to get rid of it.
Mark Twain, American author, Tom Sawyer, ch. 5, 1876

When a spring is stretched from its equilibrium length by an amount x which
is no longer small, Hooke’s law must be modified so as to include nonlinear or
anharmonic terms in x. For symmetric oscillations of a nonlinear spring about
equilibrium, the next terms retained in the Taylor expansion of the force must
be cubic in x. In this case, the force F required to stretch the nonlinear spring
will be of the form F =k1 x + k2 x3 =k1 (x + a x3), with k1 > 0 and a ≡ k2/k1.
If a > 0, one has a hard spring because it is harder to stretch than a linear
spring (a=0). If a < 0, one has a soft spring. In this recipe, we will solve the
equation of motion for the oscillations of a mass m attached to a (light) hard
spring allowed to move horizontally on a smooth horizontal surface. Applying
Newton’s second law, the equation of motion of m at arbitrary time τ is

m ẍ(τ) + k1 (x(τ) + a x(τ)3)=0 (8.2)

or, on introducing a new time variable t=
√

k1/m τ ,

ẍ(t) + x(t) + a x(t)3 =0. (8.3)

Unlike the situation for the linear spring where the solution is in terms of ele-
mentary functions (sine or cosine), the solution for the hard spring will involve
another “special” function, the Jacobian elliptic function. As you will see, by
using computer algebra, deriving the solution is not as hard as it seems.

Taking, say, a=1, the equation of motion (8.3) is entered in ode.
> restart: a:=1:

> ode:=diff(x(t),t,t)+x(t)+a*x(t)ˆ3=0;

ode := (
d2

dt2
x (t)) + x (t) + x (t)3 = 0

The dsolve command is applied to ode. Two implicit solutions are generated,
with the time t given in integral form ( a is the integration variable) and two
arbitrary constants, C1 and C2 , present. The positive square root solution
must be selected in order to produce a positive period for the oscillatory motion.

> sol:=dsolve(ode,x(t)); #choose positive root

sol :=
∫ x(t) 2√−4 a2 − 2 a4 + 4 C1

d a − t − C2 = 0,

∫ x(t)

− 2√−4 a2 − 2 a4 + 4 C1
d a − t − C2 = 0

The positive square root solution (the first one here) is selected and differenti-
ated with respect to t. This removes the constant C2 .

> eq1:=diff(sol[1],t);

eq1 :=
2 (

d

dt
x (t))√−4 x (t)2 − 2 x (t)4 + 4 C1

− 1 = 0

eq1 is solved in eq2 for dx(t)/dt, i.e., the speed.
> eq2:=solve(eq1,diff(x(t),t));

eq2 :=
1
2

√−4 x (t)2 − 2 x (t)4 + 4 C1

In order to perform the subsequent integrations, the dependent variable x(t) is
replaced in eq2 with a time-independent symbol, say y .

> eq3:=subs(x(t)=y,eq2);

eq3 :=

√
−4 y2 − 2 y4 + 4 C1

2
Let’s call the amplitude of the oscillations A. When y ≡ x(t)=A, the speed of
the mass m will be zero. This condition is expressed in eq4 .

> eq4:=subs(y=A,eq3)=0;

eq4 :=
√−4 A2 − 2 A4 + 4 C1

2
= 0

eq4 is solved for the constant C1 , which is automatically substituted into eq3
which is relabeled as eq5 .

> _C1:=solve(eq4,_C1); eq5:=eq3;

C1 := A2 +
1
2

A4 eq5 :=

√
−4 y2 − 2 y4 + 4 A2 + 2 A4

2
Since dx(t)/dt ≡ dy/dt = eq5 , the period T (time for a complete oscillation)
is given by T = 2

∫ A

−A
(1/eq5 ) dy. This integral is entered. To accomplish the

integration, the assumption that A > 0 must be included.
> T:=2*int(1/eq5,y=-A..A) assuming A>0;

T :=
4

√
2 EllipticK(

√
2 A

2
√

1 + A2
)

√
2 + 2 A2

Highlighting EllipticK in the output and opening the Help page reveals that the
period T involves the complete elliptic integral of the first kind. The incomplete
elliptic integral of the first kind is defined [AS72] by

F (φ\α)=
∫ φ

0
(1 − sin2 α sin2 θ)−1/2 dθ, (8.4)

or, on setting m ≡ sin2 α, y ≡ sin θ, and x ≡ sin φ,

F (φ |m)=
∫ x

0
[(1 − y2) (1 − m y2)]−1/2 dy. (8.5)

The complete elliptic integral K ≡ K(m) corresponds to setting φ = π/2 in
F (φ\α), or x = 1 in F (φ |m). The command EllipticF(sin(phi),sqrt(m))
generates F (φ |m), while EllipticK(sqrt(m)) produces K(m). The period is
now plotted over the range A=0 to 5, the result being shown in Figure 8.3.

> plot(T,A=0..5,labels=["A","T"],view=[0..5,0..2*Pi]);

Figure 8.3: Period T of the hard spring versus amplitude A.

Unlike the situation for a linear spring, the period of the hard spring is ampli-
tude dependent, decreasing with increasing values of A. To determine x(t), the
integral t=

∫ x

−A
(1/eq5 ) dy is evaluated assuming that A > 0 and x > −A.

> eq6:=t=int(1/eq5,y=-A..x) assuming A>0,x>-A;

eq6 := t =
2

√
2 EllipticK(

√
2 A

2
√

1 + A2
)

√
2 + 2 A2

−
√

2 EllipticF(
√

A2 − x2

A
,

√
2 A

2
√

1 + A2
)

√
2 + 2 A2

eq6 is now solved for x, which produces positive and negative square root ex-
pressions, which are assigned the names x1 and x2 . Both solution branches will
be needed to plot the complete oscillation of the hard spring.

> eq7:=solve(eq6,x); x1:=eq7[1]; x2:=eq7[2];

x1 : =√
1− JacobiSN(

1
2
(t

√
2+2A2− 2

√
2 EllipticK(

√
2A

2
√

1+A2
))

√
2,

√
2 A

2
√

1+A2
)2 A,

x2 : =

−
√

1−JacobiSN(
1
2
(t

√
2+2A2−2

√
2 EllipticK(

√
2 A

2
√

1+A2
))

√
2,

√
2 A

2
√

1+A2
)2 A

The “special” function JacobiSN appearing in the output is the Jacobi elliptic
sine function. Setting u ≡ F (φ\α), φ is referred to as the “amplitude” of u,
written as φ=am u. The elliptic sine function is defined as snu=sin(amu)=
sin φ. The Maple command for numerically calculating snu for a given m value
is evalf(JacobiSN(u,sqrt(m))). The reader is referred to Abramowitz and
Stegun [AS72] for the properties of the elliptic functions.

To plot the hard spring solution, let’s choose a specific value of the ampli-
tude, say A=3. The period T1 is first evaluated

> T1:=evalf(eval(T,A=3));

T1 := 2.294401860
and then a piecewise function x is created, using x1 for t < T1/4, x2 for
T1/4 < t < 3T1/4, and so on.

> x:=piecewise(t<T1/4,x1,t<3*T1/4,x2,t<5*T1/4,x1,

t<7*T1/4,x2,t<9*T1/4,x1,t<11*T1/4,x2):
Then x is evaluated at A=3 and plotted along with the solution 3 cos(t) to the
corresponding linear spring equation, the result being shown in Figure 8.4.

> plot([eval(x,A=3),3*cos(t)],t=0..2*Pi,color=[red,blue],

linestyle=[1,3],labels=["t","x"]);

Figure 8.4: Solid curve: hard spring; Dashed curve: linear spring (a=0).

As expected, the hard spring oscillates more rapidly than the linear spring.

8.2 Nonlinear ODEs: Graphical Methods

Two recipes are presented which illustrate how a second-order NLODE, or a
system of two first-order equations, can be graphically solved and interpreted.

8.2.1 Joe and the Van der Pol Scroll

Sometimes one likes foolish people for their folly,
better than wise people for their wisdom.
Elizabeth Gaskell, English novelist, (1810–65)

This recipe is inspired by my reminiscences of a former student in my non-
linear physics class, whose identity I will protect by calling him “Joe”. In class,

I had derived the “equation of motion” for a tunnel diode electrical circuit, the
ODE taking the form of the nonlinear Van der Pol (VdP) equation,

ẍ + ε(x2 − 1) x + x = 0, (8.6)

where x is related to the voltage change across the diode and the positive
parameter ε depends on certain circuit parameters. From a mechanical view-
point, the VdP equation follows on applying Newton’s second law to a unit
mass which experiences a Hooke’s law restoring force, FHooke =−x, and a drag
force, Fdrag =−ε (x2 − 1) ẋ. For ε=0, Eq. (8.6) reduces to the simple harmonic
oscillator equation which has undamped oscillatory solutions. For ε > 0, the
drag force has a rather peculiar property. For x > 1 and ẋ > 0, Fdrag < 0, which
tends to reduce the size of the oscillations, but for x < 1 and ẋ still positive,
Fdrag > 0, and the oscillations tend to grow in amplitude. In the tunnel diode
case, it is the latter feature which causes the diode to begin to spontaneously
oscillate even if it is connected to a steady (non-oscillatory) power supply.

As a follow up to the class room derivation, I had then asked the class
to solve the VdP equation for ε = 5 for a few initial conditions of their own
choosing, using a graphical/numerical technique called the method of isoclines.
The basis of the isoclines method is as follows. Setting y ≡ ẋ, the second-order
VdP equation can be reduced to a system of two coupled first-order ODEs, viz.,

ẋ = y, ẏ = ε (1 − x2) y − x. (8.7)

Since the time doesn’t explicitly appear1 in (8.7), it may be eliminated by form-
ing the ratio dy/dx = (ε(1−x2) y−x)/y. But this ratio is just the slope tangent
to the solution “trajectory” in the x-y plane (called the phase plane) at an in-
stant in time. A tangent field picture can be created by drawing systematically
spaced arrows in the x-y plane, the arrows oriented along the slope directions
and the arrow heads pointing in the direction of increasing time. Given some
initial condition, x(0), y(0), the solution curve (called a phase-plane trajectory)
can be drawn in the x-y plane by following the arrows. A tangent field picture
with one or more superimposed solution curves is referred to as a phase-plane
portrait of the autonomous ODE or ODE system.

The method of isoclines was commonly used in the pre-computer age to sys-
tematically draw the tangent field arrows. In this method, curves corresponding
to different constant slopes (the isoclines) were drawn and equally spaced ar-
rows pointing in the slope direction placed on each isocline. Once the x-y plane
was filled with a sufficiently fine grid of arrows, one could see how a solution
would evolve with time from any point in the phase plane.

So, why do I remember Joe? It’s because when Joe handed in his solution
to the problem (a week late!), it was in the form of a large cylindrical scroll
fastened with an elastic band. Removing the band, I began to unwind the
scroll. To my amazement, I found that the scroll stretched across the width of
my office. Joe had evidently chosen the wrong scale for his initial conditions and
stubbornly kept splicing sheets together until he obtained a complete solution

1The equations are referred to as autonomous.

trajectory. To compound Joe’s woes, he had not written a simple program to
automate the process but had evidently used a calculator and then plotted the
isoclines, arrows, and trajectories completely by hand. However, to his credit,
the plot was basically correct and I didn’t have the heart to penalize Joe for
handing the problem in late.

The following recipe carries out the solution to Joe’s problem painlessly
and quickly. It makes use of the phaseportrait command which is in Maple’s
DEtools library, so this package is loaded. I will take ε=5.

> restart: with(DEtools): epsilon:=5:

Equations (8.7) are entered in ode1 and ode2 .
> ode1:=diff(x(t),t)=y(t);

ode1 :=
d

dt
x (t) = y(t)

> ode2:=diff(y(t),t)=epsilon*(1-x(t)ˆ2)*y(t)-x(t);

ode2 :=
d

dt
y(t) = 5 (1 − x (t)2) y(t) − x (t)

Two initial conditions are considered. The phase-plane trajectory will start
from near the origin for ic1 , and reasonably far from the origin for ic2 .

> ic1:=x(0)=0.01,y(0)=0.01: ic2:=x(0)=-1,y(0)=6:

The following phaseportrait command is used to plot the tangent field2 and
the two trajectories corresponding to the above initial conditions. In the argu-
ment, the two ODEs are entered as a list, as are the dependent variables to be
solved for. The time range is taken to be t = 0 to 20 for the solution curves
and the initial conditions are entered as a list of lists. The plot range is taken
to be x = −2.5 to 2.5, and y = −10 to 13. To obtain accurate solution curves,
the numerical step size is taken to be 0.01. The dirgrid option controls the
number of tangent arrows to be drawn. Here, I have chosen to plot 30 × 30=900
arrows. The default is 20 × 20. Various arrow styles are available. I have cho-
sen arrows=MEDIUM and colored the arrows blue. Finally, the two trajectories
are colored red and green. The result is shown on the left of Figure 8.5.

> phaseportrait([ode1,ode2],[x(t),y(t)],t=0..20,[[ic1],[ic2]],

x=-2.5..2.5,y=-10..13,stepsize=.01,dirgrid=[30,30],
arrows=MEDIUM,color=blue,linecolor=[red,green]);

The tangent arrows provide a visual guide to how trajectories will evolve as the
time increases. As time evolves, the trajectory corresponding to ic1 unwinds
in a spiral fashion from near the origin onto a closed loop, indicative of a cyclic
solution. The trajectory corresponding to ic2 winds onto the same closed loop.
In fact, no matter what the initial conditions, all trajectories wind onto the
closed loop, which is an example of a limit cycle. In fact, since all trajectories
wind onto it as t → ∞, it is referred to as a stable limit cycle.

By including the option scene=[t,x(t)], the phaseportrait command
can also be used to plot x(t) versus t.

2Isoclines are not drawn.

Figure 8.5: Left: trajectories winding onto limit cycle. Right: x(t) versus t.

> phaseportrait([ode1,ode2],[x(t),y(t)],t=0..30,[[ic1],[ic2]],

x=-2.5..2.5,y=-10..13,stepsize=.01,scene=[t,x(t)],
linecolor=[red,green]);

The result is shown on the right of Figure 8.5. After a transient interval, the
two curves have identical shapes. The steady-state curves are characterized
by periods of relatively slowly varying x, periodically interspersed with abrupt
changes. These types of oscillations are referred to as relaxation oscillations.
By increasing ε, the period of the oscillations can be increased.

8.2.2 Squid Munch (Slurp?) Herring

Man is the only animal that can remain on friendly terms with the
victims he intends to eat until he eats them.
Samuel Butler, English author, (1835–1902)

Another pre-computer approach to qualitatively sketching possible solution
curves of a second-order autonomous nonlinear ODE (or two coupled first-
order NLODEs) in the phase-plane was to first locate all the stationary or
singular points (where all derivatives vanish) of the nonlinear ODE, identify
their “topological” nature, and use this knowledge to sketch the trajectories.
More specifically, this topological approach is as follows.

Consider the pair of first-order autonomous ODEs,

ẋ = P (x, y), ẏ = Q(x, y), (8.8)

where, in general, P and Q are nonlinear functions and we have taken time
as the independent variable. For the Van der Pol oscillator of the last recipe,
P ≡ y and Q ≡ −x + ε (1 − x2) y. A stationary point (x0, y0) corresponds to

ẋ = 0 and ẏ = 0, or P (x0, y0) = Q(x0, y0) = 0. For the Van der Pol oscillator,
there is only one stationary point, namely (x0 =0, y0 =0), i.e., the origin. For
any ordinary point outside a stationary point, we can write its coordinates as
x=x0 + u, y=y0 + v. The slope of the trajectory at an ordinary point is

dy

dx
=

Q(x0 + u, y0 + v)
P (x0 + u, y0 + v)

. (8.9)

For an ordinary point close to a stationary point, the numerator and denomi-
nator on the rhs can be Taylor expanded about (x0, y0) in powers of u and v, so

dy

dx
=

dv

du
=

c u + d v + · · ·
a u + b v + · · · , (8.10)

where a ≡ (∂P/∂x)x0,y0 , b ≡ (∂P/∂y)x0,y0 , c ≡ (∂Q/∂x)x0,y0 , d ≡ (∂Q/∂y)x0,y0 .
Provided that b c − a d �= 0, the trajectories near the stationary point can be
correctly described by retaining only the linear terms in u and v in (8.10). In
this case, the stationary point is referred to as simple. In this approximation,
equation (8.10) can be thought of as resulting from the pair of linear ODEs,

u̇ = a u + b v, v̇ = c u + d v, (8.11)
which have solutions of the form (u, v) ∼ eλ t, with λ=−p

2 ± 1
2

√
p2 − 4 q, where

p=−(a + d) and q=a d − b c. Detailed examination of the two λ roots reveals
that there are only four types of simple stationary points, the saddle, focal or
spiral, nodal, and vortex points. Which type occurs depends on the ranges of q,
p, and p2 − 4 q as indicated in Table 8.1.

Stationary Point q=a d–b c p=–(a+d) p2 – 4 q

saddle < 0 ≥ 0 and ≤ 0 > 0
higher order = 0 ≥ 0 and ≤ 0 ≥ 0
stable focal > 0 < 0
stable nodal > 0 ≥ 0

vortex or focal > 0 = 0 < 0
unstable focal < 0 < 0
unstable nodal < 0 ≥ 0

Table 8.1: Classification of stationary or singular points.

In the neighborhood of these simple stationary points, the trajectories have
the schematic form shown in Figure 8.6, the arrows indicating increasing time.
Stable focal and nodal points are shown, the trajectories approaching the sta-
tionary points as t → ∞. For unstable focal and nodal points, the arrow direc-
tions are reversed. The origin of the phase-plane for the Van der Pol oscillator
is an example of an unstable focal point, which can be seen in Figure 8.5.

For q > 0 and p = 0, note that either a vortex or focal point occurs. The
reason for the “uncertainty” is that inclusion of quadratic (or higher) terms in

Figure 8.6: Curves near a focal (F ), nodal (N), vortex (V ), saddle (S) point.

the Taylor expansion may turn a closed loop (for the vortex) into a spiral. In-
stead of examining these higher-order terms, we quite often rely on a sufficient,
but not necessary, global theorem due to Poincaré, which is built on symme-
try considerations. If P (x, −y) = −P (x, y) and Q(x, −y) = Q(x, y), then the
stationary point is a vortex, not a focal point.

For q = 0 and arbitrary p, the stationary point is no longer simple, and is
referred to as a higher-order stationary point. Trajectories in the neighborhood
of these points tend to be more complicated than in Figure 8.6.

In the pre-computer age, we would locate and identify the stationary points
and attempt to draw the trajectories in the entire phase-plane by splicing the
trajectories around each stationary point together. This procedure works well
enough when the nonlinear ODE has only a few widely separated simple sta-
tionary points, with no higher-order points or limit cycles present. But in this
computer age, we can simply use, e.g., the phaseportrait command. However,
it’s still important to know the locations and types of stationary points as this
provides a deeper understanding of the solution curves. This is illustrated in
the following mathematical biology example.

The major food source for squid is herring. If S and H are the numbers of
squid and herring, respectively, per acre of seabed, the interaction between the
two species can be modeled [Sco87] by the system (with time in years)

Ḣ = k1 H − k2 H2 − k3 H S, Ṡ = −k4 S − k5 S2 + k6 H S, (8.12)

with k1 = 1.1, k2 = 10−5, k3 = 10−3, k4 = 0.9, k5 = 10−4, and k6 = 2 ×
10−5. The first term on the rhs of Ḣ represents the natural growth of the
herring population if the resources (food) were unlimited, the second term limits
the growth because of the finite resources, and the third term represents the
decrease in population due to interaction (i.e., being eaten) with the squid. You
should be able to interpret the terms in the other equation.

(a) Locate and classify all the stationary points of the NLODEs. Plot the
tangent field and show the trajectories for (i) H(0) = 5, S(0) = 5; (ii)
H(0) = 10000,S(0) = 1500; (iii) H(0) = 120000,S(0) = 500. Take t = 0 to
50. Relate the results to the stationary points.

(b) Suppose that every squid was removed from the area occupied by the
herring and from all surrounding areas. Would H increase indefinitely
or would it approach an upper limit? If you believe the latter would
occur, what is that number? If the squid free situation had persisted for
many years, what are S and H two years later if a pair of fertile squid is
introduced into the area? Round the numbers to the nearest integer.

The DEtools library package is needed for the phaseportrait command.
> restart: with(plots): with(DEtools):

The coefficient values are entered,
> k1:=1.1: k2:=10ˆ(-5): k3:=10ˆ(-3): k4:=0.9: k5:=10ˆ(-4):

k6:=2*10ˆ(-5):
and functional operators P and Q introduced to generate the rhs of the NLODEs
(8.12) for arbitrary H and S.

> P:=(H,S)->k1*H-k2*Hˆ2-k3*H*S; Q:=(H,S)->-k4*S-k5*Sˆ2+k6*H*S;

P := (H, S) → k1 H − k2 H2 − k3 H S

Q := (H, S) → −k4 S − k5 S2 + k6 H S

Using P and Q, the necessary derivatives to calculate a, b, c, d are calculated.
These expressions have to be evaluated at each stationary point.

> a:=diff(P(H,S),H); b:=diff(P(H,S),S); c:=diff(Q(H,S),H);

d:=diff(Q(H,S),S);

a := 1.1 − H

50000
− S

1000
b := − H

1000
c :=

S

50000
d := −0.9 − S

5000
+

H

50000
The number of stationary points and their coordinates is determined by setting
P (H, S)=0 and Q(H, S)=0 and solving for H and S.

> sol:=solve({P(H,S)=0,Q(H,S)=0},{H,S});
sol := {H = 0., S = 0.}, {H = 110000., S = 0.}, {H = 0., S = −9000.},

{S = 619.0476190, H = 48095.23810}
The NLODEs have four stationary points at the locations indicated above. Note
that the stationary point at H = 0, S = −9000 is in a non-physical portion of
the phase-plane, since we must have H ≥ 0 and S ≥ 0.

Functional operators p and q are now formed to calculate p=−(a + d) and
q=a d − b c for the ith stationary point.

> p:=i->evalf(eval(-(a+d),sol[i])):

q:=i->evalf(eval(a*d-b*c,sol[i])):
In the following do loop, p, q, and r ≡ p2 − 4 q are calculated for each of the
four singular points. The classification scheme of Table 8.1 is implemented by
using a conditional if...then...elif...else...end if statement. elif is
a contraction of “else if”. This allows each singular point to be classified. For
this conditional statement to work, numerical (not symbolic) values must be
provided for the coefficients in the NLODEs.

> for i from 1 to 4 do

> sol[i]; p||i:=p(i); q||i:=q(i); r||i:=simplify(p||iˆ2-4*q||i);

> if q||i<0 then s||i:=saddle;

elif q||i>0 and p||i>0 and r||i<0 then s||i:=stablefocal;
elif q||i>0 and p||i>0 and r||i>=0 then s||i:=stablenodal;
elif q||i>0 and p||i<0 and r||i<0 then s||i:=unstablefocal;
elif q||i>0 and p||i<0 and r||i>=0 then s||i:=unstablenodal;
elif q||i>0 and p||i=0 then sing||i:=vortex or focal;
else s||i:=higherorder; end if: s||i;

> end do;

{H = 0., S = 0.} p1:= −0.2000000000 q1:= −0.9900000000

r1:= 4. saddle
{H = 110000., S = 0.} p2:= −0.2000000000 q2:= −1.430000000

r2:= 5.760000000 saddle
{H = 0., S = −9000.} p3:= −11. q3:= 9.090000000

r3:= 84.64000000 unstablenodal

{H = 48095.23810, S = 619.0476190} p4:= 0.5428571428 q4:= 0.6252380953

r4:= −2.206258504 stablefocal
So, the stationary points at H = 0, S = 0 and H = 110, 000, S = 0 are saddle
points, H =0, S =−9000 is an unstable nodal point, and H � 48095, S �619 is
a stable focal point. This implies that all trajectories in the “physical region”
H ≥ 0, S ≥ 0 will be attracted to the stable focal point as t → ∞. I.e., if the
equations are valid for all t (which they wouldn’t be!), one would ultimately
end up with about 48095 herring and 619 squid per acre of seabed. To confirm
this, let’s create a phase-plane portrait, first entering the NLODE system.

> sys:=diff(H(t),t)=P(H(t),S(t)),diff(S(t),t)=Q(H(t),S(t));

sys :=
d

dt
H (t) = 1.1H (t) − 1

100000
H (t)2 − 1

1000
H (t)S (t),

d

dt
S (t) = −0.9S (t) − 1

10000
S (t)2 +

1
50000

H (t)S (t)

A functional operator PP is formed to generate the phase-plane portrait S(t)
vs. H(t), as well as H(t) vs. t and S(t) vs. t, for the system of NLODEs.

> PP:=(X,Y)->phaseportrait([sys],[H(t),S(t)],t=0..50,

[[H(0)=5,S(0)=5],[H(0)=10000,S(0)=1500],
[H(0)=120000,S(0)=500]],stepsize=.05,scene=[X,Y],
arrows=MEDIUM,color=blue,dirgrid=[20,20],
linecolour=[red,green,black],xtickmarks=3):

Which plot is produced depends on the choice of scene variables X and Y. The
lists of equations and dependent variables are entered as arguments, along with
the time range which is taken here to be 0 to 50 years. The initial population
numbers are given as a list of lists. The numerical time stepsize is taken to
be 0.05 years and medium blue arrows are chosen for the tangent field. The
grid density for the tangent arrows is taken to be 20 × 20. The trajectories
corresponding to the three initial conditions are colored red, green, and black,
and the number of tickmarks along the horizontal axis controlled.

Then PP is used to produce the phase-plane portrait, as well as plots of H(t)
versus t and S(t) versus t.

> PP(H(t),S(t)); PP(t,H(t)); PP(t,S(t));

0

600

1200

S(t)

50000 100000H(t)
0

600

1200

S(t)

20 40t

Figure 8.7: Phase-plane portrait (left) and S vs. t (right) for 3 initial conditions.

S vs. H is shown on the left of Figure 8.7, and S vs. t on the right. In the
phase-plane picture on the left, one can clearly see that all three trajectories
asymptotically wind onto the stable focal point, as was anticipated. The trajec-
tory corresponding to H(0)=5, S(0)=5 initially travels horizontally very close
to the H(t) axis until it gets close to the saddle point at S =0, H =110, 000. It
then makes an abrupt turn, heading towards the stable focal point and winding
onto it. The trajectory starting at H(0)=10, 000, S(0)=1500 initially descends
almost vertically until it begins to follow the tangent field in the vicinity of the

saddle point at the origin. The fourth singular point at H =0, S =−9000 plays
no role in the “flow” of the trajectories. The phase-plane portrait doesn’t show
the times involved, but of course the other plot on the rhs of Figure 8.7 does.

If every squid was removed from the area occupied by the herring and
from all surrounding areas, H would asymptotically approach an upper limit of
110,000 herring (the H value of the saddle point to which the trajectory would
be attracted). To answer the last part of part (b), we numerically solve the
system of NLODEs with the initial condition H(0)=110, 000, S(0)=2, giving
the output as a listprocedure.

> sol2:=dsolve({sys,H(0)=110000,S(0)=2},{H(t),S(t)},
type=numeric,output=listprocedure):

The number of herring and squid at an arbitrary time t years later is now
determined using sol2.

> H:=eval(H(t),sol2): S:=eval(S(t),sol2):

The number (per acre of seabed) of herring and squid two years later is now
calculated, and (using the round command) found to be about 108,794 and 26.

> H2:=round(H(2)); S2:=round(S(2));

H2 := 108794 S2 := 26

8.3 Nonlinear ODEs: Approximate Methods

The perturbation and Krylov–Bogoliubov approximation methods, which may
be applied when the nonlinear terms in the NLODEs are small, are illustrated,
along with the Ritz trial function method which can be used when they are not.

8.3.1 Poisson’s Method Isn’t Fishy

Let a man get up and say, Behold, this is the truth, and instantly
I perceive a sandy cat filching a piece of fish in the background.
Look, you have forgotten the cat, I say.
Virginia Woolf, British novelist, (1882–1941)

Perturbation methods are applicable to NLODEs when the nonlinear terms
are small. As a simple example, consider the motion of a unit mass moving in
a viscous medium characterized by a Hooke’s law restoring force FHooke = −y
and a drag force Fdrag = −a v − ε v3 = −a (dy/dt) − ε (dy/dt)3 where y is the
displacement from equilibrium, v is the velocity, a is a positive coefficient, and
ε is a small positive parameter. The governing NLODE then is

ÿ + a ẏ + ε ẏ 3 + y = 0. (8.13)

For ε = 0, (8.13) reduces to the linearly damped simple harmonic oscillator
equation. Our goal is to derive an approximate analytic solution when ε is
small, but not zero, for the initial condition y(0)=1, ẏ(0)=0. Two values of a
will be considered, first a = 1/2 which is entered below, and later, a = 0. The
Poisson perturbation method is to assume a series solution in powers of ε, viz.,

y(t)=y0(t) + ε y1(t) + ε2 y2(t) + · · · + εN yN (t). In our recipe, let’s take N =2,
i.e., we will stop at second order in ε in our expansion.

> restart: with(plots): N:=2: a:=1/2:

A functional operator is introduced for generating the governing NLODE.
> ode:=y->diff(y(t),t,t)+a*diff(y(t),t)

+epsilon*(diff(y(t),t))ˆ3+y(t)=0;

ode := y → (
d2

dt2
y(t)) + a (

d

dt
y(t)) + ε (

d

dt
y(t))3 + y(t) = 0

The Poisson perturbation expansion is now entered out to order N .
> Y(t):=sum(y[n](t)*epsilonˆn,n=0..N);

Y (t) := y0(t) + y1(t) ε + y2(t) ε2

The perturbation expansion is automatically substituted into the NLODE in
the following command line and the result expanded. The very lengthy output
is suppressed with a line-ending colon.

> ode2:=expand(ode(Y)):

Powers of ε are now collected in ode2 , all powers of order ε(N+1) and higher
being set equal to zero. The result is then simplified.

> eq1:=simplify(collect(ode2,epsilon),{epsilonˆ(N+1)=0}):
Since ε is arbitrary (but small), the coefficient of each power of ε must be equal
to zero. This will generate a linear ODE for each order of the perturbation
expansion. These equations are produced in the following do loop.

> for n from 0 to N do

> Eq[n]:=coeff(lhs(eq1),epsilon,n)=0;

> end do;

Eq0 := (
d2

dt2
y0(t)) +

1
2

(
d

dt
y0(t)) + y0(t) = 0

Eq1 := (
d2

dt2
y1(t)) + (

d

dt
y0(t))3 + y1(t) +

1
2

(
d

dt
y1(t)) = 0

Eq2 := (
d2

dt2
y2(t)) + 3 (

d

dt
y0(t))2 (

d

dt
y1(t)) +

1
2

(
d

dt
y2(t)) + y2(t) = 0

The zeroth order equation, Eq0, is just the linearly damped simple harmonic
oscillator equation. It is now solved subject to the initial condition y0(0) = 1,
ẏ0(0) = 0, and the solution is assigned.

> sol[0]:=dsolve({Eq[0],y[0](0)=1,D(y[0])(0)=0},y[0](t));
assign(sol[0]):

sol0 := y0(t) =
1
15

√
15 e(− t

4 ) sin(
√

15 t

4
) + e(− t

4 ) cos(
√

15 t

4
)

Since the zeroth-order solution satisfies the initial condition, all higher-order
equations must be solved subject to yn(0) = ẏn(0) = 0, where n = 1, · · · , N .

This is done in the following do loop, using the Laplace transform option of the
dsolve command. Each solution is assigned and the do loop ended.

> for n from 1 to N do

> sol[n]:=dsolve({Eq[n],y[n](0)=0,D(y[n])(0)=0},y[n](t),
method=laplace); assign(sol[n]):

> end do:

The lengthy perturbation expansion is now written out3 in Y ,
> Y:=sum(’epsilonˆn*rhs(sol[n])’,’n’=0..N):

and ε and exponential terms sequentially collected.
> Y:=collect(Y,[epsilon,exp]);

Y := (
1
15

sin(
√

15 t

4
)
√

15+cos(
√

15 t

4
)) e(− t

4 )

+ε ((− 4
305

cos(
3

√
15 t

4
)+

116
13725

sin(
3

√
15 t

4
)
√

15+
4
5
cos(

√
15 t

4
)

+
4
75

sin(
√

15 t

4
)
√

15)e(− 3 t
4 )+(−48

61
cos(

√
15 t

4
)+

8
305

sin(
√

15 t

4
)
√

15) e(− t
4 ))

+ε2 ((
127494
161101

cos(
√

15 t

4
) − 10818

805505
sin(

√
15 t

4
)
√

15) e(− t
4 )

−(
108
61

cos(
√

15 t

4
) +

84
1525

sin(
√

15 t

4
)
√

15 +
32

1525
sin(

3
√

15 t

4
)
√

15) e(− 3 t
4 )

+(− 648
211975

cos(
5

√
15 t

4
) − 1544

3179625
√

15sin(
5

√
15 t

4
) − 486

28975
cos(

3
√

15 t

4
)

+
2306
86925

sin(
3

√
15 t

4
)
√

15+
28944
28975

cos(
√

15 t

4
)+

6896
86925

sin(
√

15 t

4
)
√

15)e(− 5 t
4 ))

Clearly, deriving the above solution by hand would be a tedious task. Let’s now
take a specific value for the parameter, say ε=0.4.

> epsilon:=0.4:

The perturbation solution Y is plotted over the time interval t=0 to 20 in gr1.
> gr1:=plot(Y,t=0..20,labels=["t","y"],thickness=2):

For comparison, the NLODE is now solved numerically, subject to the given
initial condition,

> X:=dsolve({ode(x),x(0)=1,D(x)(0)=0},x(t),numeric,
output=listprocedure):

and plotted in gr2. A point style is chosen, with fifty size 12 black circles.
> gr2:=odeplot(X,[t,x(t)],0..20,style=point,symbol=circle,

symbolsize=12,numpoints=50,color=black):
The two graphs are superimposed with the display command,

> display([gr1,gr2]);

3Note the enclosure of the summand and summation index in “right quotes” here. This is
required for the sum to be carried out. See Maple’s Help on sum for a discussion of this issue.

Figure 8.8: Numerical (circles) and perturbation (line) solutions for a = 1/2
(left) and a=0 (right).

the resulting picture being shown on the left of Figure 8.8. In this case, the
perturbation expansion fits the numerical solution very well.

However, if the recipe is re-executed with a = 0, trouble occurs. Y then
takes the following form, containing secular terms involving powers of t.

Y = cos(t)+ε (
1
32

sin(3 t) +
9
32

sin(t) − 3
8

t cos(t))

+ ε2(− 9
64

t sin(t)+
27
128

t2 cos(t) − 9
256

t sin(3 t) − 3
1024

cos(5 t)+
3

1024
cos(t)).

Referring to the right hand side picture in Figure 8.8, the perturbation solution
agrees with the exact numerical result for short times, but eventually the secular
terms cause the perturbation expansion to diverge from the exact solution, the
oscillations in fact increasing with time, rather than decreasing. Going to higher
order in the perturbation expansion will not “cure” the situation, as even higher
powers of t then occur. Their are a variety of methods (see Zwillinger [Zwi89])
for dealing with the presence of secular terms in the perturbation solution of a
NLODE. In the following recipe, we illustrate one method which can be applied
to a NLODE having a periodic solution.

8.3.2 Lindstedt Saves the Day

I like to do all the talking myself.
It saves time, and prevents arguments.
Oscar Wilde, Anglo-Irish playwright, author, (1854–1900)

Consider a spring whose displacement from equilibrium satisfies the NLODE,

ÿ(t) + y(t) + ε y(t) 5 = 0, (8.14)

with ε > 0 and small. Our goal is to derive a second-order perturbation solution
to equation (8.14), subject to the initial condition y(0) = 1, ẏ(0) = 0. The
solution then will be compared to the exact numerical solution for ε = 1/2. If
the Poisson method is applied, secular terms will occur, which would destroy
the expected periodic solution. So what do we do?

For ε=0, (8.14) is just the simple harmonic oscillator equation which has the
periodic solution cos(t), with frequency 1. The Lindstedt perturbation method
is to assume not only a perturbation expansion for y, but that the frequency
changes slightly to a new frequency Ω, which can also be represented by a per-
turbation series. Introducing a new time variable T =Ω t, and setting X ≡ y,
equation (8.14) can be rewritten as

Ω2 Ẍ(T ) + X(T ) + ε X(T ) 5 = 0. (8.15)

Lindstedt’s procedure is to assume that

X(T ) = x0(T )+ε x1(T )+ε2 x2(T )+· · · , and Ω = 1+ε ω1+ε2 ω2+· · · . (8.16)

The secular terms are removed in each order by imposing the periodicity condi-
tion X(T + 2 π)=X(T ), i.e., xn(T + 2 π)=xn(T ), for n=0, 1, 2, .... The initial
conditions are x0(0)=1, ẋ0 =0, and xn(0)=0, ẋn(0)=0 for n=1, 2, ....

The following recipe carries out Lindstedt’s method for the given NLODE.
The maximum power, N =2, of ε to be kept in the expansion is entered. The
following alias command produces the output symbol ε when e is entered.

> restart: with(plots): N:=2: alias(epsilon=e):

The NLODES (8.14) and (8.15) are entered in ode1a and ode1b, respectively.
> ode1a:=diff(y(t),t,t)+y(t)+e*y(t)ˆ5=0;

ode1a := (
d2

dt2
y(t)) + y(t) + ε y(t)5 = 0

> ode1b:=Omegaˆ2*diff(X(T),T,T)+X(T)+e*X(T)ˆ5=0;

ode1b := Ω2 (
d2

dT 2 X (T )) + X (T ) + εX (T )5 = 0

The perturbation expansions for X(T ) and Ω, given in (8.16), are inputted.
> X(T):=sum(’x||n(T)*eˆn’,’n’=0..N);

X (T ) := x0 (T ) + x1 (T ) ε + x2 (T ) ε2

> Omega:=1+sum(’omega||n*eˆn’,’n’=1..N);

Ω := 1 + ω1 ε + ω2 ε2

The above series are automatically substituted into ode1b, which is expanded.
> ode2:=expand(ode1b):

Powers of ε are collected in ode2 , and terms higher than εN set equal to 0.
> ode3:=simplify(collect(ode2,e),{eˆ(N+1)=0}):

A functional operator eq is formed to set the coefficient of the nth power of ε
on the lhs of ode3 equal to zero.

> eq:=n->coeff(lhs(ode3),e,n)=0:



314 CHAPTER 8. NLODES & PDES OF PHYSICS

Using the functional operator in the following sequence command produces the
linear ODEs eq0 , eq1 , and eq2 , corresponding to ε0, ε1, and ε2, respectively.

> eqs:=seq(eq||n=eq(n),n=0..N); assign(eqs);

eqs := eq0 = ((
d2

dT 2 x0 (T )) + x0 (T ) = 0),

eq1 = (x1 (T ) + (
d2

dT 2 x1 (T )) + x0 (T )5 + 2 ω1 (
d2

dT 2 x0 (T )) = 0),

eq2 = ((
d2

dT 2 x2 (T )) + 5 x0 (T )4 x1 (T ) + 2ω1 (
d2

dT 2 x1 (T )) + x2 (T )

+ ω12 (
d2

dT 2 x0 (T )) + 2ω2 (
d2

dT 2 x0 (T )) = 0)

An operator sol is created to analytically solve the nth order ODE, using the
Laplace transform method. The order n and initial value A of xn must be given.

> sol:=(n,A)->dsolve({eq||n,x||n(0)=A,D(x||n)(0)=0},x||n(T),
method=laplace):

Setting A=1, the zeroth-order ODE is solved and the solution assigned.
> sol0:=sol(0,1); assign(sol0):

sol0 := x0 (T ) = cos(T )
The 1st ODE is solved with A=0, trig terms are combined, and sol1 assigned.

> sol1:=combine(sol(1,0),trig); assign(sol1);

sol1 := x1 (T ) =
5

128
cos(3 T ) − 1

24
cos(T ) +

1
384

cos(5 T )

− 5
16

T sin(T ) + T sin(T ) ω1

The secular terms, − 5
16

T sin(T ) + T sin(T ) ω1, occur and must be removed

from x1 (T ) to maintain periodicity, i.e., to have x1 (T + 2π) = x1 (T ). This
implies that we must have ω1 = 5/16. This frequency is extracted and the
secular term removed from x1 (T ) in the following command lines.

> omega||1:=solve(coeff(x||1(T),sin(T))=0,omega||1);

x||1(T):=subs(sin(T)=0,x||1(T));

ω1 :=
5
16

x1 (T ) :=
1

384
cos(5 T ) +

5
128

cos(3 T ) − 1
24

cos(T )

Then, the second-order ODE is solved, and the solution assigned.
> sol2:=combine(sol(2,0),trig); assign(sol2):

sol2 := x2 (T ) =
1

98304
cos(9 T ) +

95
294912

cos(7 T ) − 5
192

cos(3 T )

+
3791

147456
cos(T ) + T sin(T ) ω2 +

215
3072

T sin(T )

Secular terms occur in sol2 , which are removed by choosing ω2=−215/3072.
> omega||2:=solve(coeff(x||2(T),sin(T))=0,omega||2);

x||2(T):=subs(sin(T)=0,x||2(T));

ω2 :=
−215
3072

x2 (T ) :=
1

98304
cos(9 T ) +

95
294912

cos(7 T ) − 5
192

cos(3 T ) +
3791

147456
cos(T )

The perturbation solution is now completely determined to second order in ε.
Setting ε=1/2 and T =Ω t, the Lindstedt perturbation solution X is displayed.
This would be a tedious result to derive by hand.

> e:=1/2: X:=subs(T=Omega*t,x||0(T)+e*x||1(T)+eˆ2*x||2(T));

X :=
581327
589824

cos(
13993 t

12288
) +

1
768

cos(
69965 t

12288
) +

5
384

cos(
13993 t

4096
)

+
1

393216
cos(

41979 t

4096
) +

95
1179648

cos(
97951 t

12288
)

A graph of X, as well as the ε = 0 solution, is created over the time interval
t=0 to 20.

> gr1:=plot({X,cos(t)},t=0..20):
The original NLODE, ode1a, is numerically solved and plotted over the same
time interval, size 12 black circles being used for the numerical curve.

> Y:=dsolve({ode1a,y(0)=1,D(y)(0)=0},y(t),numeric,
output=listprocedure):

> gr2:=odeplot(Y,[t,y(t)],0..20,style=point,symbol=circle,

symbolsize=12,numpoints=120,color=black):
The two graphs are superimposed, the resulting picture being shown in Fig-
ure 8.9. The Lindstedt curve fits the numerical data quite well, and differs
appreciably from the ε=0 solution.

> display({gr1,gr2});

–1

0

1

10 t 20

Figure 8.9: Lindstedt curve lies on numerical circles. Other curve: ε=0 solution.

8.3.3 Krylov–Bogoliubov Have A Say

He who says he hates every kind of flattery, and says it in earnest,
certainly does not yet know every kind of flattery.
G. C. Lichtenberg, German physicist and philosopher, (1742–99)

For a=0, equation (8.13) reduces to

ÿ + y + ε ẏ 3 = 0, (8.17)
the initial condition being y(0)=1, ẏ(0)=0. For ε small, we saw that the Poisson
perturbation expansion broke down for this NLODE, generating secular terms
which grew with time. Since the solution of this equation is not periodic, the
Lindstedt procedure cannot be used to remove the secular terms.

A procedure developed by Krylov and Bogoliubov [KB43] can be employed
to solve NLODEs of the general structure

ẍ + ω2
0 x + ε f(x, ẋ) = 0, (8.18)

with ε a small parameter. Equation (8.17) is of this structure with x = y,
ω0 =1 and f = ẋ 3. For ε=0, a general solution of (8.18) would be of the form
x=a sin(ω0 t+φ), with a and φ constants determined by the initial conditions.
If x is the displacement, the velocity would be ẋ = a ω0 cos(ω0 t + φ). In the
Krylov–Bogoliubov (KB) method4, exactly the same forms are assumed for x
and ẋ, but the amplitude a and phase factor φ are allowed to become time-
dependent. Provided that a(t) and φ(t) are “slowly varying”, their structures
are determined by solving [EM00] [Zwi89],

ȧ = − ε

2πω0

∫ 2π

0
f(a sin ψ, aω0 cos ψ) cos ψ dψ,

φ̇ =
ε

2πaω0

∫ 2π

0
f(a sin ψ, aω0 cos ψ) sin ψ dψ,

(8.19)

where ψ ≡ ω0 t + φ. The criterion for being slowly varying is that∣∣∣∣ ȧa
∣∣∣∣ 2 π

ω0
� 1, and

∣∣∣∣∣ φ̇

ω0

∣∣∣∣∣� 1. (8.20)

Recalling that ω0 = 1, for equation (8.17) we have f = (a cos ψ)3. Since the
relevant integrand cos3 ψ sin ψ is an odd function of ψ, φ̇ = 0, so φ is a constant.
We need only determine the form of a(t) and make sure that the solution
x = a(t) sin(t + φ) satisfies the initial condition. We shall now do this for
the same initial condition as in Recipe 08-3-1, taking x(0) = 1 and ẋ(0) = 0.
The KB solution will be compared to the numerical solution for the same value
of ε as in that recipe, viz., ε=0.4.

Use of the following alias command allows us to enter e in the input and
generate ε in the output.

4Also referred to as the method of averaging [Zwi89].

> restart: with(plots): alias(epsilon=e):

The form of f is entered in terms of ψ,
> f:=(a(t)*cos(psi))ˆ3;

f := a(t)3 cos(ψ)3

and the integration in the ȧ equation in (8.19) carried out.
> eq:=diff(a(t),t)=-(e/(2*Pi))*int(f*cos(psi),psi=0..2*Pi);

eq :=
d

dt
a(t) = −3

8
ε a(t)3

The resulting first-order ODE eq is analytically solved for a(t),
> sol:={dsolve(eq,a(t))};

sol := {a(t) =
2√

3 ε t + 4 C1
, a(t) = − 2√

3 ε t + 4 C1
}

generating two answers with C1 an arbitrary integration constant. The rhs
of the positive square root solution (the first one here) is selected to form the
solution x. The two constants C1 and φ remain to be determined.

> x:=rhs(sol[1])*sin(t+phi);

x :=
2 sin(t + φ)√
3 ε t + 4 C1

Taking ε=0.4, we set x(0)=1 in initial condition ic1 ,
> e:=0.4: ic1:=eval(x,t=0)=1;

ic1 :=
1
2

√
4 sin(φ)√

C1
= 1

and ẋ(0) = 0 in ic2 .
> ic2:=eval(diff(x,t),t=0)=0;

ic2 := −0.07500000000
√

4 sin(φ)

C1 (3/2) +
1
2

√
4 cos(φ)√

C1
= 0

Then ic1 and ic2 are numerically solved for φ and C1 . The solution sol2 is
assigned, and the final KB solution, x, displayed.

> sol2:=fsolve({ic1,ic2},{phi,_C1}); assign(sol2): x:=x;

sol2 := { C1 = 0.9769696007, φ = −11.14792061}

x :=
2 sin(t − 11.14792061)√

1.2 t + 3.907878403
The KB solution is plotted over the time interval t=0 to 20.

> gr1:=plot(x,t=0..20,labels=["t","x"]):

The NLODE is entered,
> ode:=diff(y(t),t,t)+y(t)+e*diff(y(t),t)ˆ3=0;

ode := (
d2

dt2
y(t)) + y(t) + 0.4 (

d

dt
y(t))3 = 0

solved numerically subject to the same initial condition,

> Y:=dsolve({ode,y(0)=1,D(y)(0)=0},y(t),numeric,
output=listprocedure):

and plotted using 50 size 12 black circles to represent the numerical curve.
> gr2:=odeplot(Y,[t,y(t)],0..20,style=point,symbol=circle,

symbolsize=12,numpoints=50,color=black):
The KB and numerical solutions are superimposed with the display command,

> display([gr1,gr2]);

the resulting picture being shown in Figure 8.10.

Figure 8.10: Circles: numerical solution. Curve: KB solution.

The KB solution is in quite good agreement with the exact (numerical) solution.

8.3.4 A Ritzy Approach

We must never confuse elegance with snobbery.
Yves Saint Laurent, French couturier, Ritz, no. 85 (London, 1984)

The perturbation and Krylov–Bogoliubov methods can be applied to a NLODE
when the nonlinear terms are small compared to the linear parts of the equation.
The Ritz method is more general, not being restricted to small nonlinearities.
Consider a general ODE,

f(x, ẋ, ẍ, ..., t) = 0, (8.21)
where f is some nonlinear function of its arguments. In the spirit of the
Rayleigh–Ritz method for estimating eigenvalues, we try to guess at a trial
wave function Φ which comes “close” to satisfying (8.21). Since, in general, Φ
will not be an exact solution, substitution of Φ into (8.21) will not generate
0 on the right-hand side, but some time-dependent contribution e(t), referred
to as the residual, i.e., f(Φ, Φ̇, Φ̈, ... t) = e(t). Paralleling the method of least
squares for fitting experimental data, the Ritz method calculates the total error

E =
∫ b

a

e2(t) dt, (8.22)

over a range a to b of interest. One then tries to choose a Φ which minimizes the
total error. As in the Rayleigh–Ritz method, one or more adjustable parameters
may be included and the error is minimized with respect to these parameters.

As an example, let’s apply the Ritz method to the hard spring equation,

ẍ + x + x3 = 0, (8.23)

with x(0)≡ A=2, ẋ(0)=0. Although we have already solved the hard spring
equation analytically in terms of Jacobian elliptic functions, the Ritz solution
will be compared to the numerical solution. As a trial function, let’s take

Φ = C1 cos(ω t) + (A − C1) cos(3 ω t), (8.24)

with C1 and ω adjustable parameters. At t=0, Φ(0)=A, Φ̇(0)=0, so the initial
conditions are satisfied.

The value of A is entered along with the upper limit T = n 2 π/|ω| in the
time range t=0 to T “of interest” that will be used for plotting purposes. For
sake of definiteness, let’s take n=4, so that T =8π/|ω|, with ω to be determined.

> restart: with(plots): A:=2: n:=4: T:=n*2*Pi/abs(omega):

The trial function Φ is entered.
> Phi:=C1*cos(omega*t)+(A-C1)*cos(3*omega*t);

Φ := C1 cos(ω t) + (2 − C1 ) cos(3 ω t)
A functional operator e for generating the residual of the NLODE (8.23) is
formed. Entering e(Phi) then will produce the residual.

> e:=x->diff(x,t,t)+x+xˆ3;

e := x → (
d2

dt2
x) + x + x3

The total error is calculated, the assumption that ω > 0 being provided to
facilitate the integration. A lengthy expression in terms of C1 and ω results.

> Error:=int(e(Phi)ˆ2,t=0..T) assuming omega>0;

Error := π(1256C1 4 + 4032 ω2 C1 − 1376C1 + 2416C1 2 + 2592 ω4

− 390C1 5 − 2304 ω2 + 55C1 6 + 544 − 2592C1 ω4 + 656C1 2 ω4

+1728C1 3 ω2 − 3712C1 2 ω2 − 2256C1 3 − 312C1 4 ω2)/(2 ω)
The total error must be minimized with respect to ω and C1 . To this end, the
derivatives of the Error with respect to ω and C1 are set equal to zero in cond1
and cond2 . The very lengthy algebraic equations are not shown here.

> cond1:=diff(Error,omega)=0; cond2:=diff(Error,C1)=0;

The two conditions are numerically solved for C1 , ω and the solution assigned.
> sol:=fsolve({cond1,cond2},{C1,omega}); assign(sol):

sol := {ω = −1.978539105, C1 = 1.937051656}

The fully evaluated total error, trial function Φ, and time T , are now displayed.

> Error:=evalf(abs(Error)); Phi:=Phi; T:=evalf(T);

Error := 0.2146672855

Φ := 1.937051656 cos(1.978539105 t) + 0.062948344 cos(5.935617315 t)

T := 12.70267601

The trial function Φ is plotted over the time interval t=0 to T . Since Φ is our
“best” approximation to the exact solution x, the ordinate is labeled as x.

> gr1:=plot(Phi,t=0..T,labels=["t","x"]):

The NLODE is solved numerically, subject to the same initial condition,

> Y:=dsolve({e(y(t))=0,y(0)=A,D(y)(0)=0},y(t),numeric,
output=listprocedure):

and plotted over the same time range. A point style is chosen, the numerical
curve being represented by 150 size 12 black circles.

> gr2:=odeplot(Y,[t,y(t)],0..T,style=point,symbol=circle,

symbolsize=12,numpoints=150,color=black):
The graphs of the Ritz and numerical solutions are superimposed,

> display([gr1,gr2]);

the resulting picture being shown in Figure 8.11. The Ritz solution is in excel-

Figure 8.11: Circles: numerical solution. Curve: Ritz solution.

lent agreement with the numerical points over the time range of the plot. It
should be mentioned that there are several variations on the Ritz approximation
method, which are discussed in Zwillinger’s book.

8.4 Nonlinear PDEs

Most of the nonlinear PDEs that my graduate students and I have had to solve
in our research in nonlinear optics and fluid dynamics have involved either
seeking special solutions, or more often using numerical methods. Here, a few
examples of the former are presented, the latter being covered in Chapter 9.

8.4.1 John Scott Russell’s Chance Interview

The boisterous sea of liberty is never without a wave.
Thomas Jefferson, American president, (1743–1826)

Shortly after Jefferson’s death, the great Scottish naval architect, John Scott
Russell, made the following important scientific observation:

I was observing the motion of a boat which was rapidly drawn along a
narrow channel by a pair of horses, when the boat suddenly stopped –
not so the mass of water in the channel which it had put in motion;
it accumulated round the prow of the vessel in a state of violent
agitation, then suddenly leaving it behind, rolled forward with great
velocity, assuming the form of a large solitary elevation, a rounded
smooth and well-defined heap of water, which continued its course
along the channel apparently without change of form or diminution
of speed. I followed it on horseback, and overtook it still rolling on
at a rate of some eight or nine miles an hour, preserving its original
figure some thirty feet long and a foot to a foot and a half in height.
Its height gradually diminished, and after a chase of one or two
miles I lost it in the windings of the channel. Such, in the month of
August 1834, was my first chance interview with that singular and
beautiful phenomenon . . .

The relevant nonlinear PDE describing the water waves for the above situa-
tion was derived several decades later by Korteweg and deVries, the Korteweg-
deVries (KdV) equation taking the form [KdV95]

∂ψ

∂t
+ ψ

∂ψ

∂x
+

∂3ψ

∂x3 = 0, (8.25)

with x the distance, t the time, and ψ the transverse displacement of the water
from equilibrium. Damping of the water waves is completely neglected here.
What Scott Russell observed was a special localized solution of the KdV equa-
tion, referred to as a solitary wave solution. The form of the solitary wave may
be obtained by assuming that ψ(x, t)=U(z = x−c t), subject to the asymptotic
boundary conditions that U(z) and all its derivatives vanish as |z| → ∞. The
parameter c will be the speed of the solitary wave. In this recipe, U will be
derived and the solitary wave animated for two different values of c.

The plots and PDEtools packages are loaded, the former containing the
animation command, while the PDEtools package has the dchange command
which will be used to transform the variables in the KdV equation.

> restart: with(plots): with(PDEtools):

The KdV equation (8.25) is entered.
> pde:=diff(psi(x,t),t)+psi(x,t)*diff(psi(x,t),x)

+diff(psi(x,t),x,x,x)=0;

pde := (
∂

∂t
ψ(x, t)) + ψ(x, t) (

∂

∂x
ψ(x, t)) + (

∂3

∂x3 ψ(x, t)) = 0

The variable transformation x = z + c τ , t = τ , and ψ(x, t) = U(z) is entered,
x, t and ψ(x, t) being the “old” variables, while z, τ , and U(z) are the “new”
variables. This transformation will have the effect of reducing the nonlinear
pde to a nonlinear ODE.

> tr:={x=z+c*tau,t=tau,psi(x,t)=U(z)};
tr := {x = z + c τ, t = τ, ψ(x, t) = U (z)}

The variable transformation tr is applied to pde using the dchange command.
> ode1:=dchange(tr,pde,[z,tau,U(z)]);

ode1 := −(
d

dz
U (z)) c + U (z) (

d

dz
U (z)) + (

d3

dz3 U (z)) = 0

The lhs of ode1 is now integrated. Since U(z) and its second derivative must
vanish at z=∞, the integration constant is equal to zero. The integrated result
is set equal to zero, yielding the second order NLODE ode2 .

> ode2:=int(lhs(ode1),z)=0;

ode2 := −cU (z) +
1
2

U (z)2 + (
d2

dz2 U (z)) = 0

Then ode2 is analytically solved for U(z), yielding two implicit solutions.
> sol:=dsolve(ode2,U(z));

sol :=
∫ U (z) 3√−3 a3 + 9 c a2 + 9 C1

d a − z − C2 = 0,

∫ U (z)

− 3√−3 a3 + 9 c a2 + 9 C1
d a − z − C2 = 0

The positive square root solution (the first one here) is now differentiated with
respect to z, and dU/dz is isolated to the lhs of the slope equation.

> slope:=isolate(diff(sol[1],z),diff(U(z),z));

slope :=
d

dz
U (z) =

1
3

√−3U (z)3 + 9 cU (z)2 + 9 C1

Since U(z) and dU(z)/dz vanish at z=∞, the integration constant C1 can be
set equal to zero. The radical simplification command is also applied in ode3 .

> ode3:=radsimp(eval(slope,_C1=0));

ode3 :=
d

dz
U (z) =

1
3

√
3U (z)

√−U (z) + 3 c

The first order ODE ode3 is analytically solved.
> sol2:=dsolve(ode3,U(z));

sol2 := z +
2 arctanh(

1
3

√−3U (z) + 9 c√
c

)
√

c
+ C1 = 0

Then sol2 is solved for U(z). The integration constant C1 determines the
location of the peak of the solitary wave and can be set equal to zero without
loss of generality.

> U:=eval(solve(sol2,U(z)),_C1=0);

U := 3 c − 3 tanh(
z

√
c

2
)2 c

Converting U to a sine/cosine form and simplifying with the trig option gives
us the final form of the solitary wave solution in terms of the variable z.

> U:=simplify(convert(U,sincos),trig);

U :=
3 c

cosh(
z

√
c

2
)2

U has a maximum height at z = 0 which is proportional to c, a width which
decreases with increasing c, and goes to zero as |z| → ∞. We can convert back
to the original variables by substituting z=x−c t into U . ψ, given in the output
of the following command line, is the solitary wave observed so long ago by John
Scott Russell. This is clearly a localized pulse which travels unchanged in shape
in the positive x direction with velocity c. Since the height is proportional to
c, this implies that taller KdV solitary waves travel faster than shorter ones.
We can confirm this by animating the solitary wave profile for two different
velocities. First, let’s apply the unapply command to ψ, turning it into a
functional operator f depending on the value of c.

> psi:=subs(z=x-c*t,U); f:=unapply(psi,c):

ψ :=
3 c

cosh(
(x − c t)

√
c

2
)2

Taking c = 1 and c = 3, two solitary waves are animated by entering f(1) and
f(3) in the animate command. On executing the following command line on
the computer, and clicking on the plot, the animation will begin.

> animate({f(1),f(3)},x=-10..40,t=0..10,numpoints=200,
frames=50,thickness=2,axes=framed,labels=["x","psi"]);

The initial profile of the animation is shown in Figure 8.12. As time progresses,
the shorter solitary wave lags behind the taller one.

Figure 8.12: Initial frame of the solitary wave animation.

This difference in speeds between solitary waves of different amplitudes will be
used in the next chapter to numerically test their stability against collisions
with one another. If they survive unchanged, except perhaps for a phase shift,
they are called solitons.

8.4.2 There is a Similarity

Ought not education to bring out and fortify the differences
rather than the similarities?
Virginia Woolf, British novelist, referring to the gender issue, (1882–1941).

In the previous recipe, the number of independent variables was reduced from
two to one, thus converting a nonlinear PDE to an ODE for which a physically
important exact solution could be obtained. It was a simple application of the
similarity method. Now I will show you a slightly more complicated similarity
solution, solving a nonlinear diffusion equation by taking a different algebraic
combination of the independent variables.

Consider the one-dimensional nonlinear diffusion equation,
∂C

∂t
=

∂

∂x

(
D(C)

∂C

∂x

)
, D(C) = Cn, (8.26)

with the diffusion coefficient D no longer constant, but a function of the con-
centration C. Several special cases of (8.26) have been studied in the litera-
ture, viz., n = 3 to model the spreading of thin liquid films under the action
of gravity [Buc77], n ≥ 1 to describe the percolation of gas through porous
media [Mus37], and n=6 to model certain radiative heat transfer [LP80].

After loading the PDEtools and plots library packages,
> restart: with(PDEtools): with(plots):

the nonlinear diffusion equation (8.26) is entered. The inert form of the deriva-
tive operator is used to prevent the rhs from being explicitly differentiated.

> pde:=Diff(C(x,t),t)-Diff(C(x,t)ˆn*diff(C(x,t),x),x)=0;

pde := (
∂

∂t
C (x, t)) − (

∂

∂x
(C (x, t)n (

∂

∂x
C (x, t)))) = 0

New variables z, τ , and U(z) are introduced, related to the old variables x, t,
and C(x, t) by the transformation x=z τm, t=τ , and C(x, t)=U(z)/τm, with
the parameter m to be determined.

> tr:={x=z*tauˆm,t=tau,C(x,t)=U(z)/tauˆm};

tr := {x = z τm, t = τ, C (x, t) =
U (z)
τm

}
The transformation is applied to pde using the dchange command, the result
then being multiplied by −τm+1 and expanded.

> ode:=expand(-tauˆ(m+1)*dchange(tr,pde,[z,tau,U(z)]));

ode := (
d

dz
U (z)) z m + U (z) m +

τ (
U (z)
τm

)n n (
d

dz
U (z))2

(τm)2 U (z)

+
τ (

U (z)
τm

)n (
d2

dz2 U (z))

(τm)2
= 0

To remove all τ factors from ode and thus really produce an ODE, we must
have (τm)2 =τ (1/τm)n which is now entered.

> eq:=(tauˆm)ˆ2=tau*(1/tauˆm)ˆn;

eq := (τm)2 = τ (
1

τm
)n

Then eq is solved for m, yielding m=1/(n + 2). This m value will be automat-
ically substituted into ode.

> m:=solve(eq,m);

m :=
1

2 + n
Dividing ode by m, simplifying symbolically with the power option, and then
applying the general simplification command, yields the formidable looking
NLODE given in ode2

> ode2:=simplify(simplify(ode/m,power,symbolic));

ode2 := (
d

dz
U (z)) z + U (z) + 2U (z)(n−1) n (

d

dz
U (z))2 + U (z)n (

d2

dz2 U (z)) n

+ U (z)(n−1) n2 (
d

dz
U (z))2 + 2U (z)n (

d2

dz2 U (z)) = 0

A general analytic solution to ode2 is sought using the dsolve command, the
result then being simplified symbolically with the power option.

> sol:=simplify(dsolve(ode2,U(z)),power,symbolic);



326 CHAPTER 8. NLODES & PDES OF PHYSICS

sol := U (z) = b( a) &where

[{ b( a)n (
d

d a
b( a)) +

a b( a) + 2 C1 + C1 n

2 + n
= 0},

{ a = z, b( a) = U (z)}, {z = a, U (z) = b( a)}]
A general solution for U(z) is not produced in sol , but U(z)= b( a) where b( a)
satisfies a first order NLODE. A solution can be generated if the integration
constant C1 is set equal to zero. The following operand command is used to
extract the resulting ODE from sol .

> _C1:=0: ode3:=op([2,2,1],sol)[1];

ode3 := b( a)n (
d

d a
b( a)) +

a b( a)
2 + n

= 0

Then, ode3 is analytically solved for b( a).

> sol2:=dsolve(ode3,_b(_a));

sol2 := b( a) =
(
− a2 n + 4 C2 + 2 C2 n

2 + n
)(

1
n )

2( 1
n )

The integration constant C2 controls the height and width of the solution.
Here, I will take C2=1. Substituting a =z on the rhs of sol2 yields U(z).

> U:=subs({_a=z,_C2=1},rhs(sol2));

U :=
(
−z2 n + 4 + 2n

2 + n
)(

1
n )

2( 1
n )

The above solution only exists for z values such that (−z2 n+4+2n)/(2+n) ≥ 0.
Outside this range, we can take U = 0, which clearly satisfies the NLODE.
Substituting z =x/tm into U/tm gives us a solution, labeled c, in terms of the
original variables.

> c:=subs(z=x/tˆm,U/tˆm);

c :=

⎛
⎜⎜⎜⎝

− x2 n

(t(
1

2+n ))2
+ 4 + 2n

2 + n

⎞
⎟⎟⎟⎠

( 1
n )

2( 1
n ) t(

1
2+n )

The x range for which c is valid is found by setting c=0 and solving for x. The
result is simplified with the radical simplification command and labeled X.

> X:=radsimp(solve(c=0,x));

X :=
√

2
√

n (2 + n) t(
1

2+n )

n
As a specific example, let’s take n=3, so the solution can be used to describe
the spreading of a thin liquid film under the action of gravity. Other n values

can be chosen, if so desired. The total time range is taken to be T =1500, and
X is evaluated at this time and assigned the name X0 .

> n:=3: T:=1500: X0:=eval(X,t=T):

The complete concentration profile C is then described by the following piece-
wise function.

> C:=piecewise(abs(x)<X,c,abs(x)>X,0);

C :=

⎧⎪⎪⎪⎨
⎪⎪⎪⎩

2(2/3) (− 3 x2

5 t(2/5) + 2)(1/3)

2 t(1/5) |x| <

√
2

√
15 t(1/5)

3

0
√

2
√

15 t(1/5)

3
< |x|

A 3-dimensional plot of C is now produced,

> plot3d(C,x=-X-5..X+5,t=1..T,numpoints=1500,axes=box,labels

=["x","t","C"],orientation=[40,70],tickmarks=[3,3,3]);
the resulting picture being shown in Figure 8.13.

Figure 8.13: Similarity solution of the nonlinear diffusion equation for n=3.

The solution captures some of the experimentally observed features of spreading
thin films. Unlike the situation for the linear (D constant) diffusion equation,
there is now a sharp interface separating the regions of nonzero and zero con-
centration. Further, the interface propagates with finite speed, in contrast with
the infinite speed of linear diffusion. The propagation of this interface can be
observed by executing the following animate command, clicking on the resulting
computer plot, and on the start arrow.

> animate(C,x=-X0..X0,t=1..T,frames=50,numpoints=500,

labels=["x","C"]);
For more on similarity methods, you are referred to Bluman and Cole [BC74].

8.4.3 Creating Something Out Of Nothing

Say nothing good of yourself, you will be distrusted;
say nothing bad of yourself, you will be taken at your word.
Joseph Roux, French priest, writer, (1834–86)

The sine-Gordon equation (SGE),

∂2u

∂x2 − ∂2u

∂t2
= sinu, (8.27)

is a model equation for describing the motion of a Bloch domain wall between
two ferromagnetic domains. A solitary wave solution to the SGE, describing the
Bloch wall motion, could be obtained in a similar manner to that for the KdV
equation. An alternate way is to make use of an auto-Bäcklund transformation.
Given a solution of a nonlinear PDE, such a transformation allows us to find a
different solution of the same PDE. More, generally a Bäcklund transformation
may enable one to use the solution of one nonlinear PDE to determine the
solution of another nonlinear PDE. Bäcklund transformations are difficult to
find, so in this recipe I will merely confirm the auto-Bäcklund transformation
for the SGE and use it to create a non-trivial solitary wave solution of (8.27),
starting with the trivial null (u=0) solution.

After loading the PDEtools and plots library packages,
> restart: with(PDEtools): with(plots):

the sine-Gordon equation is entered.
> sge:=diff(u(x,t),x,x)-diff(u(x,t),t,t)=sin(u(x,t));

sge := (
∂2

∂x2 u(x, t)) − (
∂2

∂t2
u(x, t)) = sin(u(x, t))

Introducing the variable transformation x=X +T , t=X −T , u(x, t)=U(X, T ),
> tr:={x=X+T,t=X-T,u(x,t)=U(X,T)};

tr := {u(x, t) = U (X, T ), t = X − T, x = X + T}
and using the dchange command, the SGE takes the form shown in sge2 .

> sge2:=dchange(tr,sge,[X,T,U(X,T)]);

sge2 :=
∂2

∂X ∂T
U (X, T ) = sin(U (X, T ))

From Zwillinger [Zwi89], the auto-Bäcklund transformation for the SGE is given
by ab1 and ab2 , where a is an arbitrary parameter.

> ab1:=diff(V(X,T),X)=diff(U(X,T),X)+2*a*sin((V(X,T)+U(X,T))/2);

ab1 :=
∂

∂X
V (X,T ) = (

∂

∂X
U (X,T )) + 2 a sin(

1
2

V (X,T ) +
1
2

U (X,T ))

> ab2:=diff(V(X,T),T)=-diff(U(X,T),T)+(2/a)*sin((V(X,T)-U(X,T))/2);

ab2 :=
∂

∂T
V (X,T ) = −(

∂

∂T
U (X,T )) +

2 sin(
1
2

V (X,T ) − 1
2

U (X,T ))

a

If U is a solution of the SGE, and V satisfies the transformation, then V is also
a solution of the SGE. Let us confirm the transformation. First, differentiating
ab1 and ab2 with respect to T and X, respectively, and subtracting, yields 0
since ∂2V/∂T∂X =∂2V/∂X∂T .

> pde1:=diff(ab1,T)-diff(ab2,X);

Substituting ab1 and ab2 into pde1/2, and applying the combine command
with the trig option, we find that U satisfies the transformed SGE, sge2 .

> sge3:=combine(subs({ab1,ab2},pde1/2),trig);

sge3 := 0 = (
∂2

∂X ∂T
U(X, T )) − sin(U(X, T ))

Similarly, let’s differentiate ab1 and ab2 with respect to T and X, respectively,
and add the results.

> pde2:=diff(ab1,T)+diff(ab2,X);

Then, on dividing both the lhs and rhs of pde2 by 2, substituting ab1 and ab2
into the latter and applying the combine command with the trig option, we
find that V also satisfies the transformed SGE.

> sge4:=lhs(pde2)/2=combine(subs({ab1,ab2},rhs(pde2))/2,trig);

sge4 :=
∂2

∂X ∂T
V (X, T ) = sin(V (X, T ))

Having confirmed the auto-Bäcklund transformation, let’s set U(X, T ) = 0 and
calculate ab1 + a2ab2 in pde3 .

> U(X,T):=0: pde3:=ab1+aˆ2*ab2;

pde3 := (
∂

∂X
V(X, T )) + a2 (

∂

∂T
V(X, T )) = 4 a sin(

1
2

V(X, T ))

Using the pdsolve command, a general solution V of pde3 is obtained (not
displayed here), expressed in terms of the arctangent function.

> V:=rhs(pdsolve(pde3,V(X,T)));

The solution is put into a simpler form by using the operand command on V .
Here F1(T − a2 X) is an arbitrary function of the argument.

> V:=2*arctan(op([2,2],V)/op([2,1],V));

V := −2 arctan

(
1
2

e(4 X a+4 F1(T−a2 X) a) − 1
e(2 X a+2 F1(T−a2 X) a)

)

Setting T = (x − t)/2 and X = (x + t)/2 transforms us back to the original
independent variables. As an example, let’s take F1(T − a2 X)=T − a2 X.

> T:=(x-t)/2: X:=(x+t)/2: _F1(T-aˆ2*X):=T-aˆ2*X:

Then, on simplifying, V takes the form shown below.
> V:=simplify(V); s:=unapply(V,a):

V := −2 arctan(
1
2

(e(−2 a (−2 x+a2 x+a2 t)) − 1) e(a (−2 x+a2 x+a2 t)))

Having used the unapply command to change V into a functional operator s in
terms of the parameter a, which controls the solitary wave velocity, the solution

is animated for a=−0.9. The opening frame is shown on the left of Figure 8.14,
the profile changing in a localized region from V =−π to +π.

> animate(s(-0.9),x=-10..10,t=0..20,frames=100,

numpoints=500,thickness=2,labels=["x","V"]);

–3

3

V

–10 10x

–3

3

V

–10 10x

Figure 8.14: Left: sine-Gordon kink soliton. Right: Anti-kink soliton.

This profile is another example of a solitary wave solution (a “kink” solitary
wave), propagating in the positive x direction with constant shape and velocity
in the animation. It can be demonstrated that this solitary wave is collisionally
stable so is referred to as a sine-Gordon kink soliton. If we take a=0.9 in the
recipe, an anti-kink soliton solution results as shown on the right of Figure 8.14.

8.4.4 Portrait of a Nerve Impulse

Is it a fact – or have I dreamt it – that, by means of electricity,
the world of matter has become a great nerve,
vibrating thousands of miles in a breathless point of time?
Nathaniel Hawthorne, American author, (1804–64)

A graphical technique for determining whether or not a nonlinear PDE has
a solitary wave solution is now illustrated. One of the most important classes
of systems to display nonlinear diffusion is found in the nerve fibers of animals.
During the first half of the nineteenth century, physiologists assumed nervous
activity to be propagated with the speed of light. Helmholtz [Hel50] decided
to make a direct measurement of the nerve signal propagation speed in a frog’s
sciatic nerve, and obtained a speed of 32 meters per second. Nonlinear diffusion
was not understood at that time and Helmholtz conjectured that the relatively
small velocity was due to motion of the material particles. The correct un-
derstanding of what was going on eluded neurophysiologists until well into the
twentieth century. Attributing the result to A.F. Huxley, Nagumo [NYA65]



8.4. NONLINEAR PDES 331

reported a solitary wave solution to the following nerve fiber equation,
∂2φ

∂x2 =
∂φ

∂t
+ φ (φ − A) (φ − 1), (8.28)

with A a constant, and the solitary wave having a velocity V = (1 − 2 A)/
√

2.
Equation (8.28) is just the linear diffusion equation to which a cubic nonlinear
term has been added for the electric potential φ. In the absence of the nonlinear
term, the diffusion velocity is infinite.

By assuming that φ(x, t) = X(z) with z = x − V t and creating a phase-
plane portrait, Y =dX/dz vs. X(z), we shall graphically see that an anti-kink
solitary wave solution to (8.28) does exist. This graphical approach has also
been successfully used by one of my former graduate students to establish the
existence of solitary wave solutions to a very formidable nonlinear optical PDE
where an analytical solution was not possible.5

Loading the DEtools library package, let’s take, say A = 1/4 and calculate
the velocity V given by Nagumo.

> restart: with(DEtools):

> A:=1/4: V:=(1-2*A)/sqrt(2);

V :=
√

2
4

Assuming φ(x, t)=X(z = x − V t), eq. (8.28) reduces to the pair of ODEs,

X ′(z) = P (X, Y ) = Y, Y ′(z) = Q(X, Y ) = −V Y + X (X − A) (X − 1).

Functional operators P and Q are formed to calculate the rhs of these ODEs.
> P:=(X,Y)->Y; Q:=(X,Y)->-V*Y+X*(X-A)*(X-1);

P := (X, Y ) → Y Q := (X, Y ) → −V Y + X (X − A) (X − 1)

The derivatives a=
∂P

∂X
, b=

∂P

∂Y
, c=

∂Q

∂X
, and d=

∂Q

∂Y
are carried out.

> a:=diff(P(X,Y),X); b:=diff(P(X,Y),Y); c:=diff(Q(X,Y),X);

d:=diff(Q(X,Y),Y);
Setting P (X, Y )=0 and Q(X, Y )=0 and solving for X and Y , yields

> sol:=solve({P(X,Y)=0,Q(X,Y)=0},{X,Y});

sol := {Y = 0, X = 0}, {Y = 0, X =
1
4
}, {Y = 0, X = 1}

three stationary points at (X = 0, Y = 0), (1/4, 0) and (1, 0). Operators for
calculating p=−(a+d) and q=a d−b c for the ith stationary point are created.

> p:=i->evalf(eval(-(a+d),sol[i])):

q:=i->evalf(eval(a*d-b*c,sol[i])):
For each stationary point, the following do loop evaluates p, q, and r = p2 −
4 q, and, based on the values of these quantities, identifies the nature of the
stationary point.

5J.P. Ogilvie, Nonlocal Solitons in Photorefractive Materials, M. Sc. thesis, Simon Fraser
University, 1996.

> for i from 1 to 3 do

> sol[i]; p||i:=p(i); q||i:=q(i); r||i:=simplify(p||iˆ2-4*q||i);

> if q||i<0 then s||i:=saddle;

elif q||i>0 and p||i>0 and r||i<0 then s||i:=stablefocal;
elif q||i>0 and p||i>0 and r||i>=0 then s||i:=stablenodal;
elif q||i>0 and p||i<0 and r||i<0 then s||i:=unstablefocal;
elif q||i>0 and p||i<0 and r||i>=0 then s||i:=unstablenodal;
elif q||i>0 and p||i=0 then sing||i:=vortex or focal;
else s||i:=higherorder; end if: s||i;

> end do;

{Y = 0, X = 0} p1 := 0.3535533905 q1 := −0.2500000000

r1 := 1 .125000000 saddle

{Y = 0, X =
1
4
} p2 := 0.3535533905 q2 := 0.1875000000

r2 := −0.6250000001 stablefocal

{Y = 0, X = 1} p3 := 0.3535533905 q3 := −0.7500000000

r3 := 3.125000000 saddle
So, (0, 0) and (1, 0) are saddle points, while (1/4, 0) is a stable focal point. To
create the phase-plane portrait, the system of ODEs is now entered.

> sys:=diff(X(z),z)=P(X(z),Y(z)),diff(Y(z),z)=Q(X(z),Y(z));

sys :=
d

dz
X(z) = Y(z),

d

dz
Y(z) = −1

4
√

2 Y(z) + X(z) (X(z) − 1
4
) (X(z) − 1)

Instead of using the phaseportrait command to create the tangent field with
trajectories corresponding to specified initial conditions, we can form a func-
tional operator gr using DEplot to accomplish the same objective. The DEplot
command has the advantage that the X and Y ranges for the plot can be inde-
pendently fixed. An initial condition is chosen close to the stationary point at
(1,0). The corresponding trajectory will be approximately that corresponding
to the solitary wave. For other A values, this number will have to be altered if
you wish to draw the trajectory corresponding to the solitary wave. The scene
variables have to be specified.

> gr:=(x,y)->DEplot([sys],[X(z),Y(z)],z=0..25,

[[X(0)=0.9999,Y(0)=-0.0001]],X=-0.1..1.1,Y=-0.25..0.25,
stepsize=.01,scene=[x,y],arrows=MEDIUM,color=black,
dirgrid=[30,30],linecolour=red):

Making use of the above graphing operator, the phase-plane portrait (X vs. Y )
is produced along with X vs. z for the given initial condition.

> gr(X(z),Y(z)); gr(z,X(z));

The resulting pictures are shown in Figure 8.15. The behavior of the tangent
field in the phase-plane portrait shown on the left is clearly consistent with
the stationary point analysis given earlier. The trajectory corresponding to the

Figure 8.15: Left: Phase-plane portrait with saddle points at (0, 0), (1, 0), and
stable focal point at (1/4, 0). Right: X vs. z for separatrix trajectory.

chosen initial condition is approximately that of the separatrix from the saddle
point at (1, 0) to the saddle point at the origin. The separatrix separates two
distinct types of “flow”. Above the separatrix, the tangent arrows wind onto
the stable focal point at (1/4, 0), while below the separatrix, the arrows flow
off to infinity. The shape of X(z) for the separatrix trajectory is shown on the
right of the figure, clearly approximating that of an anti-kink solitary wave.

For larger values of A, the nature of the stationary points changes, but a
separatrix (solitary wave) solution still exists. The initial condition needed to
plot the separatrix must be altered accordingly.

8.5 Supplementary Recipes
08-S01: A Bunch of Bernoulli equations
For each of the following ODEs, confirm that it is a Bernoulli equation, analyt-
ically solve the ODE for y(0) = 1, and plot y(x) over a suitable range.

(a) y ′ + x2 y = x4 y3;

(b) y − y ′ = 3 y3e−2x;

(c) y3 + 3 y2y ′ = 4.

08-S02: The Riccati Equation
Another first-order nonlinear ODE which can be analytically solved is Riccati’s
equation which has the general structure

y ′ + a y2 + f1(x) y + f2(x) = 0.
(a) By introducing a new dependent variable z = e

a
∫ x

0
y(X) dX , show that

Riccati’s equation can be reduced to the linear ODE

z ′′ + f1(x) z ′ + a f2(x) z = 0,
and the solution to Riccati’s equation given by y=z ′/(a z).



334 CHAPTER 8. NLODES & PDES OF PHYSICS

(b) Taking f1 =1/x and f2 = 1/a, use the transformation in part (a) to solve
Riccati’s equation. Identify the functions which result.

(c) Solve Riccati’s equation with f1 = 1/x and f2 = 1/a directly using the
dsolve command. Making use of the infolevel command, show that
Maple identifies the ODE as of the Riccati type.

08-S03: Period of the Plane Pendulum
The equation of motion of the undamped plane pendulum is θ̈ + ω2 sin θ = 0,
where the angle θ is measured from the vertical. Analytically determine the
period T of the plane pendulum for arbitrary angular amplitude A, with |A| <
π. Taking ω=1, plot T versus A over the range A=0 to π. Discuss the result.

08-S04: The Child–Langmuir Law
In a vacuum diode, electrons (each of charge q and mass m) are emitted from a
hot planar cathode at x=0, held at zero potential, and accelerated across a gap
to a parallel planar anode at x=d, held at a positive potential V0. The cloud of
moving electrons within the gap (called the space charge) quickly builds up to
the point where it reduces the electric field at the cathode to zero. Thereafter, a
steady current I flows between the plates. If the cathode and anode plate areas
A >> d, edge effects can be neglected, and the potential V between the plates is
governed by the 1-dimensional Poisson equation, d2V (x)/dx2 =−ρ(x)/ε0, where
ρ(x) is the electron charge density and ε0 is the permittivity of free space.
(a) Assuming that the electrons start from rest at the cathode and that

steady-state prevails, determine V (x) between the plates. Plot V (x) along
with the form that the potential would have without space charge.

(b) Show that the current I satisfies the Child–Langmuir law I = K V
3/2
0 ,

where the constant K remains to be determined in terms of V0, A, q, m,
d, and ε0. With space charge present, a nonlinear relation exists between
the current and voltage, i.e., Ohm’s law is not satisfied.

08-S05: Soft Spring
The displacement x(t) of a soft spring from equilibrium satisfies the NLODE

ẍ + ω2
0 (1 − b2 x2) x = 0,

with ω0 =1 and b=2. Create a phase-plane portrait showing the tangent field
in the y ≡ ẋ vs. x plane and trajectories for the initial conditions (x(0) = 0.4,
y(0)=0.01), (x(0)=−0.941, y(0)=0.9), and (x(0)=0.941, y(0)=−0.9). Take
the time range to be t=0 to 20, the time stepsize to be 0.01, the grid to be 30
by 30, and the x and y ranges to each be from −1 to +1. Plot x vs t for the
three trajectories, and discuss the results.

08-S06: Gnits vs. Gnots
The gnits (N1 per unit area) and gnots (N2 per unit area) are competing for
the same food supply, and are described by the following system of NLODES,

Ṅ1 = (a1 − b1 N1 − c1 N2) N1, Ṅ2 = (a2 − b2 N1 − c2 N2) N2,

with a1 =4, b1 =0.0002, c1 =0.0004, a2 =1.4, b2 =0.00015, and c2 =0.00005.

(a) Find and identify the stationary points of this system.

(b) Produce a phase-plane portrait which includes all four stationary points,
shows the tangent field, and the approximate separatrix trajectories which
divide the phase plane into different possible outcomes for the gnit and
gnot populations. What are these possible outcomes?

(c) Plot N1 vs. t and N2 vs. t for the above trajectories.

08-S07: The Vibrating Eardrum
In the era before the introduction of computer algebra into the teaching of my
nonlinear physics class, I used to assign the following problem and would find
that a substantial fraction of the students would make algebraic mistakes and
get the wrong answer for the general case. This no longer happens in this
computer algebra age.

Consider the equation of motion of the vibrating eardrum,

ẍ + ω2
0 x + ε x2 = 0,

subject to the initial conditions x(0) = A0, ẋ(0) = 0. Using the Lindstedt
perturbation method, determine the perturbation solution to second order in
the parameter ε. Taking ω0 = 2, ε = 1, and A0 = 1/3 plot the perturbation
solution along with the numerical solution and discuss the results.

08-S08: Van der Pol Transient Growth
Taking ε to be small, and assuming x(0) = A and ẋ(0) = 0, derive the KB
solution to the Van der Pol equation. Plot the KB and numerical solutions
together for A=1 and ε=1/10 over the time interval 0 to 60, using circles to
represent the numerical curve and a solid line for the KB solution.

08-S09: Another Ritzy Solution
Derive a KB solution to the nonlinear diode equation ẋ + x + x5 = 0, with
x(0)=1. Graphically compare the KB solution with the numerical answer. You
should experiment with the form of the trial wave function until you obtain a
small total error and hence a good fit of the curves.

08-S10: Portrait of a Dark Soliton
The nonlinear Schrödinger equation,

i
∂E

∂z
± 1

2
∂2E

∂τ2 + |E|2 E = 0,

can be used to describe picosecond duration optical envelope solitons in trans-
parent optical fibers governed by the Kerr refractive index n = n0 + n2|φ|2,
with n0 > 0, n2 > 0. Here E is proportional to the electric field amplitude
φ, z to the distance coordinate Z in the direction of propagation, and τ to
t − Z/vg, where t is the time and vg is the group velocity. The plus-minus
sign corresponds to anomalous or normal dispersion, respectively. In terms
of the light intensity, which is proportional to |E|2, so-called “bright” solitons

(the intensity is a localized peak against a zero intensity background) occur for
the plus sign, while “dark” solitons (the intensity is a localized dip to zero in
a background of constant non-zero intensity) occur for the minus sign. The
solitary wave solutions are found by assuming a “stationary” solution of the
form E(z, τ) = X(τ) ei β z with X real and β real and positive. Taking β = 1,
determine the nature of the stationary points for the dark case and create a
phase-plane diagram of X(τ) versus Y = dX/dτ showing the tangent field and
the two separatrixes which divide the phase plane into regions of qualitatively
different flows. Noting that the intensity is proportional to X(τ)2, show that
the two separatrixes correspond to dark solitary waves. It can be shown that
these solitary waves are collisionally stable, and therefore are solitons.

08-S11: Bright Soliton Solution
Referring to 08-S10, determine the analytic form X(τ) of the bright solitary
wave solution and plot the intensity X2 versus τ for β =1. The bright solitary
wave solution can be shown to be collisionally stable, so is a bright soliton.



Chapter 9

Numerical Methods

To this point, when a numerical solution to an ODE was sought, we simply
resorted to the numeric option of Maple’s dsolve command. In the first section
of this chapter, we will briefly explore the underlying principles on which the
numerical algorithms for solving linear and nonlinear ODEs are based. In the
second section, some numerical recipes for solving linear and nonlinear PDEs
are presented. If you wish to learn more about numerical methods, consult,
e.g., Numerical Analysis [BF89] or Numerical Recipes[PFTV90].

9.1 Ordinary Differential Equations

As noted earlier, the second-order Van der Pol equation

ẍ − ε (1 − x2) ẋ + x = 0, ε > 0, (9.1)
can be written as the following pair of first-order ODEs,

ẋ = y, ẏ = ε (1 − x2) y − x ≡ F (x, y). (9.2)
More generally, an nth-order ODE can be decomposed into a system of n first-
order equations. Numerical algorithms for solving such a system are based on
replacing each first derivative and each function (e.g.,F ) on the rhs with some
finite difference approximation based on a finite Taylor series expansion.

For small h, y(x ± h) can be Taylor expanded in powers of h about x, viz.,

y(x ± h) = y(x) ± hy ′(x) +
1
2!

h2y ′′(x) ± 1
3!

h3y ′′′(x) + · · · (9.3)

where y ′ ≡ dy/dx, etc. For time-dependent problems, x is replaced with t.
Taking the + sign in (9.3), and dropping terms of order h2 and higher, yields
the forward difference approximation (FWDA) to y ′(x),

y ′(x) = (1/h)[y(x + h) − y(x)] + O(h), (9.4)
while the minus sign yields the backward difference approximation (BWDA),

y ′(x) = (1/h)[y(x) − y(x − h)] + O(h). (9.5)
Explicit (implicit) numerical schemes are based on the FWDA (BWDA).

9.1.1 Joe’s Problem Revisited

One man’s remorse is another man’s reminiscence.
Ogden Nash, American poet, A Clean Conscience Never Relaxes, 1938

Recall that in the tale of Joe and the Van der Pol Scroll, the phaseportrait
command was used to easily plot the solution of the VdP equation. In the
present recipe, the VdP equation is again solved, but now using Euler’s method,
the simplest of the explicit schemes. The FWDA is used for the time deriva-
tives in (9.2), thus connecting, say, the kth time step to step k + 1, and the
right-hand sides are approximated by their values at the kth step. Explicitly,
on setting tk+1 = tk + h, the Euler algorithm for numerically solving (9.2) is

xk+1 = xk + h yk, yk+1 = yk + hF (xk, yk). (9.6)

For given values of h, x0, and y0, equations (9.6) are iterated forward in time.
So, let’s begin our recipe. The plots package is loaded because we shall be

using the pointplot and odeplot commands to plot the numerical solution.
> restart: with(plots):

18 digits accuracy is specified, so a comparison can be made at a specific time
with Maple’s “dial-up” version of the Euler method which uses 18 digits. To
ensure a reasonable approximation of the Euler algorithm to the exact ODEs,
a small value of h, h=0.01 s, is chosen. n=6000 iterations are considered, so
the total time is 0.01 × 6000 = 60 s.

> Digits:=18: h:=0.01: n:=6000:

At t0 =0, let’s take x0 =0.2, y0 =0.1, and choose ε=20.
> t[0]:=0: x[0]:=0.2: y[0]:=0.1: epsilon:=20:

An operator is formed to calculate the function F .
> F:=(x,y)->epsilon*(1-xˆ2)*y-x:

An important aspect in comparing numerical algorithms is to know how long it
takes the algorithm to execute. Here, the time will be short, but let’s record it
anyways. The time() command records the total cpu time since the beginning
of the Maple session. Here it marks the beginning of our Euler iteration.

> begin:=time():

The iterative process is carried out with the following do loop. In the body of
the loop, we have the time increment, tk+1 = tk +h, followed by equations (9.6).
The triplet of numbers, tk, xk, yk on the kth time step are grouped as a Maple
list in ptk for 3-dimensional plotting purposes, and the loop is ended.

> for k from 0 to n do

> t[k+1]:=t[k]+h;

> x[k+1]:=x[k]+h*y[k];

> y[k+1]:=y[k]+h*F(x[k],y[k]);

> pt[k]:=[t[k],x[k],y[k]]:

> end do:

The time at the end of the do loop is recorded and the beginning time subtracted
to give us the elapsed cpu time for the algorithm.

> cpu_time:=time()-begin;

cpu time := 0.980
It took1 about 1 s to generate all the data points needed for plotting. A 3-
dimensional plot of t vs. x vs. y is produced, using the sequence of n data
points, with the pointplot3d command. The orientation [-90,0] produces x vs.
t as shown on the left of Figure 9.1. Each cross represents a data point. In the
slowly varying regions, the data points are so thick that they blend into a solid
line. If you wish to see the phase plane portrait, y vs. x, simply rotate the
computer plot by clicking on it and dragging with your mouse.

> pointplot3d([seq(pt[j],j=0..n)],axes=normal,symbol=cross,

symbolsize=8,color=red,labels=["t","x","y"],
orientation=[-90,0]);

–2

–1

1

2

x

20 40 t 60

–2

–1

1

2

x

20 40 t 60

Figure 9.1: Euler solution of VdP eq. Left:“First principles”; Right:“Dial-up”.

Maple has built-in options in its dsolve command to implement some common
specific numerical algorithms, the (forward) Euler method being one of them.
Let’s now illustrate this “dial-up” approach, entering the system of first-order
ODEs for the VdP oscillator.

> sys:=diff(X(t),t)=Y(t),diff(Y(t),t)=F(X(t),Y(t));

sys :=
d

dt
X (t) = Y (t),

d

dt
Y (t) = 20 (1 − X (t)2)Y (t) − X (t)

The dependent variables are specified along with the initial condition.
> vars:={X(t),Y(t)}: ic:=X(0)=0.2,Y(0)=0.1:

1All cpu times quoted in this chapter are for a 1 GHz personal computer.

The dsolve command is applied to sys, subject to the initial condition, a numer-
ical solution being sought with stepsize h. To obtain the Euler approximation,
rather than Maple’s default scheme, the option method=classical[foreuler]
is specified. The output is given as a listprocedure, so that the numerical data
can be extracted for comparison with the first-principles calculation.

> sol:=dsolve({sys,ic},vars,numeric,stepsize=h,
method=classical[foreuler],output=listprocedure):

The numerical solution, X(t) vs. t, is then plotted over the time interval t=0 to
hn = 60, using the odeplot command. The same number of points is taken as
in the first-principles calculation. The result is shown on the right of Figure 9.1,
a line style having been selected.

> odeplot(sol,[t,X(t)],0..h*n,axes=normal,style=line,

numpoints=n,labels=["t","x"]);
The two pictures appear to be identical, but we can confirm the agreement
quantitatively by considering a particular time, say, t = 1. The solution sol is
used to evaluate X at arbitrary time t in XX. Using XX, the dial-up value of x at
t=1 is given in x1 . The first-principles answer is found by entering pt[100],
the x value at 0.01 × 100 = 1.00 s being the second entry in the output.

> XX:=eval(X(t),sol): x1:=XX(1); pt[100];

x1 := 1.64892899170276830
[1.00, 1.64892899170276830, −0.0478758403582165596]

The dial-up and first-principles values of x agree to 18 digits!
Because ε was increased from 5 in our earlier recipe (08-2-1) to 20 here, the

relaxation oscillations in Figure 9.1 are characterized by even longer periods of
relative inaction between the abrupt temporal changes. It should be mentioned
that an example of a relaxation oscillator in nature (although not governed by
the VdP equation) is the famous geyser Old Faithful in Yellowstone Park.

9.1.2 Survival of the Fittest

The price which society pays for the law of competition ... is great.
But, whether the law be benign or not, ... it is here; we cannot evade
it; no substitutes for it have been found; and while the law may be
sometimes hard for the individual, it is best for the race, because it
ensures the survival of the fittest in every department.
Andrew Carnegie, American industrialist, philanthropist, (1835–1919)

In terms of a Taylor series expansion, the Euler method is of O(h) accurate.
This necessitates taking h very small, which leads to a longer computation time
than for an algorithm which is of higher order accuracy in h, thus permitting a
bigger h to be used. In this recipe, the Volterra–Lotka (V-L) equations will be
solved with the modified Euler method, which is of O(h2) accuracy, as well as
with the less accurate Euler method.

The V-L competition equations are used in mathematical biology to describe
the interaction of biological populations, such as, e.g., rats and ferrets. Letting
r(t) and f(t) be the rat and ferret population numbers per unit area at time t,
the V-L equations are

ṙ = R(r, f) = a1 r − a2 rf, ḟ = F (r, f) = −b1 f + b2 rf, (9.7)

where the coefficients a1, a2, b1, and b2 are all positive. In (9.7), it is assumed
that the ferrets survive by eating only rats, which munch on garbage. If the
rats and ferrets did not interact (a2 = b2 = 0), the rat (ferret) equation would
become ṙ = a1 r (ḟ = −b1 f), which has an exponentially growing (decaying)
solution for the rat (ferret) number. Inclusion of the nonlinear interaction terms
in each ODE will dramatically change the nature of the solution.

Setting tk+1 = tk + h, the modified Euler algorithm for (9.7) is

rk+1 = rk + h (R1[k] + R2[k])/2, fk+1 = fk + h (F1[k] + F2[k])/2, (9.8)

with R1[k] ≡ R(rk, fk), F1[k] ≡ F (rk, fk), R2[k] ≡ R(rk + hR1[k], fk + hF1[k]),
and F2[k] ≡ F (rk + hR1[k], fk + hF1[k]).

That this scheme is of O(h2) accuracy can be confirmed by expanding (9.8)
in powers of h and showing that the resulting equations agree to order h2 with
the Taylor expansions of rk+1 = r(tk+1 = tk + h), fk+1 = f(tk+1).

As an illustrative example, let’s take the nominal values a1 = 2, a2 = 0.02,
b1 =1, b2 =0.01, with r0 =300 and f0 =100 at t0 =0.

> restart: with(plots):

> a1:=2: a2:=0.02: b1:=1: b2:=0.01:

> t[0]:=0: r[0]:=300: f[0]:=100:

10 digits accuracy is taken and for the modified Euler method the stepsize
h=0.06 and n=500 iterations are considered. For the Euler method, a smaller
stepsize H = h/3 = 0.02 is chosen and a correspondingly larger number N =
3 n=1500 iterations taken to make the total time for both solutions the same.

> Digits:=10: h:=0.06: n:=500: H:=h/3: N:=3*n:

Functional operators are formed to calculate R and F , i.e., the rhs of eqs. (9.7).
> R:=(r,f)->a1*r-a2*r*f: F:=(r,f)->-b1*f+b2*r*f:

Using spacecurve, an operator gr is formed to plot the numerical solution for
either method as a red curve in 3 dimensions. The numerical points p must be
specified as well as the upper limit n in the sequence command. The orientation
is taken to be [0,90], so that a phase-plane portrait, f vs. r, results. Of course,
the 3-dimensional plots can be rotated on the computer screen.

> gr:=(p,n)->spacecurve([seq(p[k],k=0..n)],color=red,axes=

normal,labels=["t","r","f"],orientation=[0,90]):
The Euler solution will be obtained first. The start time for the algorithm is
recorded in begin1,

> begin1:=time():

and the Euler method applied to the V-L equations, the stepsize being H and
N the number of iterations .

> for k from 0 to N do

> t[k+1]:=t[k]+H;

> r[k+1]:=r[k]+H*R(r[k],f[k]);

> f[k+1]:=f[k]+H*F(r[k],f[k]);

> pt1[k]:=[t[k],r[k],f[k]];

> end do:

The cpu time for the above algorithm is determined.
> cpu_time1:=time()-begin1;

cpu time1 := 0.208
It took about 0.2 seconds cpu time to execute the Euler method. Using gr and
the numerical points pt1, the Euler solution of the V-L equations is plotted,
the resulting picture being shown on the left of Figure 9.2.

> gr(pt1,N);

Figure 9.2: Left: Euler solution. Right: Modified Euler solution.

From this picture, one might conclude that the trajectory is an unstable spiral,
unwinding from the starting point r0 = 300, f0 = 100. Let’s see what happens
when the modified Euler method is now applied to the V-L equations. Again
the starting and ending times for the algorithm are recorded.

> begin2:=time():

> for k from 0 to n do

> t[k+1]:=t[k]+h;

> R1[k]:=R(r[k],f[k]); F1[k]:=F(r[k],f[k]);

> R2[k]:=R(r[k]+h*R1[k],f[k]+h*F1[k]);

> F2[k]:=F(r[k]+h*R1[k],f[k]+h*F1[k]);

> r[k+1]:=r[k]+h*(R1[k]+R2[k])/2;

> f[k+1]:=f[k]+h*(F1[k]+F2[k])/2;

> pt2[k]:=[t[k],r[k],f[k]];

> end do:

> cpu_time2:=time()-begin2;

cpu time2 := 0.164
Because a larger stepsize was used with fewer iterations, the cpu time for the
modified Euler method is slightly less here than for the Euler algorithm. Again
using gr, and considering the numerical points pt2, the resulting trajectory is
as shown on the right of Figure 9.2.

> gr(pt2,n);

Because the cumulative numerical error for the modified Euler method is less,
a more accurate curve is obtained. It appears that the correct solution is a
cyclic variation in the rat and ferret population numbers. This is in dramatic
contrast to the situation where there is no interaction. Although the parameter
values were artificial and there are limitations2 to the V-L equations, such cyclic
behavior has been observed in a wide variety of competing biological species.

Since the modified Euler and Euler algorithms are actually equivalent to
finite Taylor expansions in h, it would not be surprising that the algorithms
break down for large values of h. What may surprise you is that the breakdown
can occur for values of h considerably less than 1, the breakdown showing up as
a numerical instability with the solution diverging to infinity. Without changing
any of the other parameters you should see what happens in this recipe as you
increase the value of h. You may have to cut down on n to avoid numerical
overflow. Numerical instabilities for sufficiently large h are an inherent feature
of explicit schemes.

9.1.3 A Chemical Reaction

What a man calls his “conscience” is merely the mental action that
follows a sentimental reaction after too much wine or love.
Helen Rowland, American journalist, A Guide to Men, 1922

The Euler and modified Euler methods are the simplest examples of single-step
explicit methods for numerically solving ordinary differential equations. They
represent the lowest approximations in terms of accuracy of the Rungé–Kutta
(RK) methods ([LS71][PFTV90]). Historically, the fourth-order (accurate to or-
der h4) RK method has been the “work horse” of these methods as it combines
reasonable accuracy with reasonable speed.

2For example, if there were no interaction, the rat population would not grow indefinitely
if confined to a given area because of limited resources, diseases, use of poisons, etc. Further,
ferrets eat other creatures such as rabbits.

Given the general ODE, dx/dt=f(t, x), the 4th-order RK algorithm is

xk+1 = xk + [K1 + 2 K2 + 2 K3 + K4]/6, (9.9)

with K1 =hf(tk, xk), K2 =hf(tk+h/2, xk+K1/2), K3 =hf(tk+h/2, xk+K2/2),
and K4 =hf(tk+h, xk+K3). The algorithm is easily generalized to higher-order
ODEs or systems of first-order ODEs.

As a simple example, consider the irreversible chemical reaction

2 K2Cr2O7 + 2 H2O + 3 S → 4 KOH + 2 Cr2O3 + 3 SO2

with initially N1 molecules of potassium dichromate (K2Cr2O7), N2 molecules
of water (H2O), and N3 atoms of sulphur (S). The number x of potassium hy-
droxide (KOH) molecules at time t seconds is given by the rate equation

dx/dt = k(2N1 − x)2(2N2 − x)2(4N3/3 − x)3

with k = 1.64 × 10−20 s−1 and x(0) = 0.
Taking N1 = 1000, N2 = 2000, N3 = 5000, stepsize h = 0.001 s, and n = 200

iterations, we will solve the rate equation using the fourth-order RK algorithm
and graphically compare the solution with that obtained using Maple’s dsolve
command with the option method=classical[rk4]. We will also determine
the number of KOH molecules at 0.1 s.

After loading the plots package and setting the accuracy at 10 digits,
> restart: with(plots): Digits:=10:

an operator F is created for calculating the rhs of the rate equation.
> F:=x->k*((2*N1-x)ˆ2)*((2*N2-x)ˆ2)*((4*N3/3-x)ˆ3);

F := x → k (2N1 − x)2 (2N2 − x)2 (
4N3

3
− x)3

The various parameter values are entered in the next two command lines.
> k:=1.64*10ˆ(-20): h:=0.001: n:=200:

> x[0]:=0: t[0]:=0: N1:=1000: N2:=2000: N3:=5000:

After recording the start time, the 4th-order RK algorithm is applied to the
rate equation, making use of F . A plotting point, pt, is formed on each step.

> begin:=time():

> for j from 0 to n do

> K1[j]:=h*F(x[j]);

> K2[j]:=h*F(x[j]+K1[j]/2);

> K3[j]:=h*F(x[j]+K2[j]/2);

> K4[j]:=h*F(x[j]+K3[j]);

> x[j+1]:=x[j]+(K1[j]+2*K2[j]+2*K3[j]+K4[j])/6;

> t[j+1]:=t[j]+h;

> pt[j]:=[t[j],x[j]];

> end do:

The cpu time for executing the algorithm is determined, and a graph of the
numerical points created in gr1. The points are represented by size 12 blue
circles. To avoid a messy overlap, only every third point is plotted.

> cpu_time:=time()-begin;

cpu time := 0.153

> gr1:=pointplot([seq(pt[3*j],j=0..n/3)],symbol=circle,

symbolsize=12,color=blue):
To determine the number of KOH molecules at 0.1 s, the plotting point at
n/2 = 100 is determined and the round command applied to the second entry.

> Pt:=pt[n/2]; N:=round(Pt[2]);

Pt := [0.100, 1586.746962] N := 1587
At 0.1 s, there are 1587 KOH molecules. Now the rate equation is solved using
the dsolve command. The relevant ODE is entered, again making use of F .

> ode:=diff(X(t),t)=F(X(t));

Taking the same stepsize h and initial condition X(0) = 0, ode is numerically
solved for X(t), with the option method=classical[rk4].

> sol:=dsolve({ode,X(0)=0},X(t),numeric,stepsize=h,
method=classical[rk4]):

This solution is then plotted in gr2 with odeplot, using a line style.

> gr2:=odeplot(sol,[t,X(t)],0..h*n,numpoints=n,style=line):

The two graphs, gr1 and gr2, are then displayed in Figure 9.3, the circles lying
exactly on top of the solid line as expected.

> display({gr1,gr2},labels=["t","x"]);

Figure 9.3: Circles: “First principles” solution. Line: “Dial-up” solution.

9.1.4 Parametric Excitation

It is the unknown that excites the ardor of scholars, who,
in the known alone, would shrivel up with boredom.
Wallace Stevens, American poet, (1879–1955)

In numerically solving some nonlinear ODEs, regions of solution space can
be encountered where the solution is changing relatively slowly and a larger
stepsize could be used while maintaining a reasonable accuracy, while in other
regions the solution is rapidly changing and a smaller stepsize should be used.
In such situations, an algorithm which adapts its stepsize according to the so-
lution “terrain” is desirable. One of the most popular adaptive step schemes is
the Rungé–Kutta–Fehlberg [Feh70] fourth-fifth (RKF 45) method, which uses
both the fourth- and fifth-order RK algorithms and adapts the stepsize accord-
ing to the difference between the two solutions. Fortunately, one of the options
in the numerical dsolve command is precisely this scheme. To invoke it, simply
include the option method=rkf45.

As an illustrative example of using the rkf45 option, consider the follow-
ing interesting problem. The pivot point for the simple pendulum (a mass m
attached to a light rod of length r making an angle θ with the vertical) is under-
going vertical oscillations given by A sin(ωt). Using the Lagrangian approach,
show that the relevant equation of motion is

θ̈ + [W 2 − A ω2 sin(ω t)/r] sin θ = 0,

with W =
√

g/r, where g is the gravitational acceleration. This nonlinear
ODE with a time-dependent coefficient is an example of a parametric excitation.
Taking A=1 m, r= 9.8 m, g=9.8 m/s2, θ(0)=2π/3 rads, and θ̇(0)=0 rads/s,
numerically solve the equation of motion using the RKF 45 method for ω=0.1
and 1.2 rads/s. Plot θ(t) over the time interval t=100 to 300 s.

The plots and VariationalCalculus packages are loaded,

> restart: with(plots): with(VariationalCalculus):

and the horizontal (x) and vertical (y) coordinates of m entered. The coordi-
nates are measured from the equilibrium position of the suspended mass.

> x:=r*sin(theta(t)); y:=r*(1-cos(theta(t)))+A*sin(omega*t);

x := r sin(θ(t)) y := r (1 − cos(θ(t))) + A sin(ω t)
The kinetic energy T = 1

2m(ẋ2 + ẏ2) is calculated and simplified.

> T:=simplify((m/2)*(diff(x,t)ˆ2+diff(y,t)ˆ2));

T :=
1
2
m(r2 (

d

dt
θ(t))2 + 2 r sin(θ(t)) (

d

dt
θ(t)) A cos(ω t) ω + A2 cos(ω t)2 ω2)

The potential energy V =m g y is entered and the Lagrangian L=T −V formed.

> V:=m*g*y: L:=T-V;

L :=
1
2
m(r2 (

d

dt
θ(t))2 + 2 r sin(θ(t)) (

d

dt
θ(t)) A cos(ω t) ω

+ A2 cos(ω t)2 ω2) − m g (r (1 − cos(θ(t))) + A sin(ω t))
The Euler–Lagrange command is applied to L, simplified, and equated to 0.

> eq:=simplify(EulerLagrange(L,t,theta(t))[1])=0;

eq := −m r (g sin(θ(t)) + r (
d2

dt2
θ(t)) − sin(θ(t)) A sin(ω t) ω2) = 0

The output in eq is the equation of motion. Because L contained an explicit
time-dependence, no first integral was produced here. Multiplying eq by −1
and dividing by m r2 produces the result in eq2 .

> eq2:=expand(-eq/(m*rˆ2));

eq2 :=
g sin(θ(t))

r
+ (

d2

dt2
θ(t)) − sin(θ(t)) A sin(ω t) ω2

r
= 0

Next, we substitute g=r W 2 into eq2 .
> eq3:=subs(g=r*Wˆ2,eq2);

and obtain the desired NLODE by collecting the sin(θ(t)) terms in eq3 .
> eq4:=collect(eq3,sin(theta(t)));

eq4 := (W 2 − A sin(ω t) ω2

r
) sin(θ(t)) + (

d2

dt2
θ(t)) = 0

The given parameter values are entered and the frequency W =
√

g/r calculated.
> A:=1: r:=9.8: g:=9.8: W:=sqrt(g/r);

W := 1.000000000
The initial condition is entered.

> ic:=theta(0)=2*Pi/3,D(theta)(0)=0;

ic := θ(0) =
2 π

3
, D(θ)(0) = 0

Applying the unapply command to eq4 turns the ODE into a functional oper-
ator, where the frequency ω must be supplied.

> ode:=unapply(eq4,omega):

An operator sol is formed for numerically solving the ODE for θ(t), using the
RKF 45 method, for a given ω and subject to the initial condition.

> sol:=omega->dsolve({ode(omega),ic},theta(t),numeric,
method=rkf45):

An operator p is created to plot the numerical solution (θ(t) vs. t) for a specified
value of ω. The time range is taken from t=100 to 300, the lower limit being
such as to eliminate the transient. To avoid an incorrect set of figures, 3000
plotting points are chosen.

> p:=omega->odeplot(sol(omega),[[t,theta(t)]],100..300,

numpoints=3000,thickness=2):

Entering p(0.1) and p(1.2) yields the pictures on the left and right of Fig. 9.4.

> p(0.1); p(1.2);

Figure 9.4: RKF 45 solution for ω=0.1 (left) and ω=1.2 (right).

For ω =0.1, a periodic response of the pendulum is obtained, while for ω =1.2
the solution appears to be highly irregular or chaotic. Other values of ω yield
different results.

9.1.5 A Stiff System

A stiff apology is a second insult ....
The injured party does not want to be compensated because he has
been wronged; he wants to be healed because he has been hurt.
G. K. Chesterton, British author, (1874–1936)

A stiff ODE, or ODE system, is one for which there are two or more very
different time or spatial scales for the independent variable. The shortest char-
acteristic time τ (or distance) acts as an approximate boundary between nu-
merical stability and instability for any explicit numerical scheme with a fixed
stepsize h. When h is less than τ , the scheme is stable, but as h is increased
above τ the scheme becomes increasingly unstable, with floating point overflow
eventually occuring. Numerical algorithms having an adaptive stepsize, such
as RKF45, circumvent this problem by adjusting the stepsize appropriately. In
this recipe, we shall flesh these ideas out by numerically solving a stiff system
of linear ODEs with both the fourth-order RK and the RKF 45 algorithms.
This system will also be solved analytically so as to illustrate the existence of
two widely different characteristic times and to provide us with a check on the
accuracy of our numerical algorithms.

Consider the following pair of coupled first-order linear ODEs with time-
dependent forcing terms present in each equation,

ẋ = 9x + 24 y + 5 cos t − 1
3

sin t, ẏ = −24 x − 51 y − 9 cos t +
1
3

sin t.

Suppose that the initial condition is x(0) = 4/3, y(0) = 2/3. Let’s begin the
recipe by entering the ODE system and the initial condition.

> restart: with(plots):

> sys:=diff(x(t),t)=9*x(t)+24*y(t)+5*cos(t)-(1/3)*sin(t),

diff(y(t),t)=-24*x(t)-51*y(t)-9*cos(t)+(1/3)*sin(t);

sys :=
d

dt
x (t) = 9 x (t) + 24 y(t) + 5 cos(t) − 1

3
sin(t),

d

dt
y(t) = −24 x (t) − 51 y(t) − 9 cos(t) +

1
3

sin(t)

> ic:=x(0)=4/3,y(0)=2/3:

The analytic forms of x(t) and y(t) follow on applying the dsolve command to
sys, subject to the initial condition.

> sol:=dsolve({sys,ic},{x(t),y(t)});

sol := {x (t) = −e(−39 t)+2 e(−3 t)+
1
3
cos(t), y(t) = 2 e(−39 t)−e(−3 t)− 1

3
cos(t)}

There are two widely differing characteristic times in the transient part of the
solution, viz., τ1 = 1/39 � 0.03 and τ2 = 1/3 � 0.3. The shorter time τ1 will
set an approximate boundary between the stability or the lack thereof of the
RK4 scheme. For later comparison, let’s plot the exact solution, first using the
following operator F to select the forms of x and y separately from sol .

> F:=v->rhs(select(has,sol,v(t))[1]):

Then, using F with v=x and y, the exact analytic solution is plotted, using the
spacecurve command, in the 3-dimensional t vs. x vs. y space.

> gr:=spacecurve([t,F(x),F(y)],t=0..20,numpoints=2000):

An operator sol2 is formed to numerically solve the ODE system, for a specified
stepsize h, using the fourth-order RK algorithm.

> sol2:=h->dsolve({sys,ic},{x(t),y(t)},numeric,stepsize=h,
method=classical[rk4]):

An operator G is created to plot the numerical solution s for a specified h and
upper time limit T . The latter is included because for h >> τ1, but less than
τ2, numerical overflow will occur if we try to plot the RK4 solution for large T .

> G:=(s,h,T)->odeplot(s(h),[t,x(t),y(t)],t=0..T,numpoints=2000,

style=line,axes=box,labels=["t","x","y"]):
Using G, the RK4 numerical solution is plotted for h = 0.05 and T = 20 and
superimposed on the exact solution (plotted in gr) with the display command.

> display({gr,G(sol2,0.05,20)});

Figure 9.5: Left: Exact & RK4 curves (h= .05). Right: RK4 & RKF45 (h= .08).

The resulting picture is shown on the left of Figure 9.5, the exact solution being
the completely smooth curve, the RK4 numerical solution deviating away from
the exact result during the transient interval but locking onto the steady-state
oscillatory solution for larger times. The slight deviation of the RK4 solution
away from the exact curve is a precursor of the onset of numerical instability
which occurs if h is increased much further. Before showing what happens, let’s
form another operator sol3 to numerically solve the ODE system for a given h
using the RKF 45 algorithm.

> sol3:=h->dsolve({sys,ic},{x(t),y(t)},numeric,method=rkf45):
Now the time stepsize h is increased to 0.08, which is still substantially less
than the other characteristic time τ2 � 0.3 in the problem. The RK4 and RKF
45 solutions for h = 0.08 are now superimposed with the display command,
the resulting curves being shown on the right of Figure 9.5.

> display({G(sol2,0.08,0.21),G(sol3,0.08,20)});
The jagged diverging curve is the RK4 solution, the smooth one the RKF 45
result. For the RKF 45 solution, derived with sol3, the upper time limit is
still 20, but for the RK4 solution, derived with sol2, T has been shortened to
0.21 to avoid floating point overflow. Noting that the x and y scales of the two
viewing boxes in Figure 9.5 are different, the RKF 45 result agrees with the
analytical result.

It should be mentioned that other numerical algorithm options are available
for solving stiff ODE systems. The reader should consult Maple’s Help under
the topic heading “dsolve,numeric” for information about these options.

9.1.6 A Strange Attractor

Wickedness is a myth invented by good people to account for the
curious attractiveness of others.
Oscar Wilde, Anglo-Irish playwright, author, (1854–1900)

Implicit numerical algorithms, based on replacing the exact first derivative with
the BWDA, are stable for any fixed stepsize h, although they may not be too ac-
curate if h is made too large. For nonlinear ODEs, the implicit schemes involve
solving simultaneous nonlinear algebraic equations at each step. In practice,
the algorithms are often made semi-implicit by linearizing the algebraic equa-
tions. In this recipe, a semi-implicit scheme is developed for the Rössler system
[Ros76], which is famous for its “top-hat” strange attractor.

The Rössler equations are

ẋ = −(y + z) ≡ X, ẏ = x + a y ≡ Y, ż = b + z (x − c) ≡ Z, (9.10)

with x, y, and z real and a, b, c real, positive, constants. They were introduced
by Rössler to illustrate how a simple 3-dimensional ODE system can exhibit
periodic and chaotic solutions, depending on the parameter values.

Using the BWDA in the Euler method, with k replaced by k+1, (9.10) yields

xk+1 = xk + h Xk+1, yk+1 = yk + hYk+1, zk+1 = zk + hZk+1. (9.11)

This resembles the Euler algorithm that was used before, except now the rhs
of (9.10) are evaluated at the “new” time step instead of the “old” one. This
is an example of an implicit scheme. If all the ODEs are linear, it is easy to
solve the equations, expressing xk+1, yk+1, and zk+1 completely in terms of
the old known values. For the Rössler system, Zk+1 contains the nonlinear
term, zk+1 xk+1, involving the new unknown values. It is customary to con-
vert implicit schemes to semi-implicit ones by Taylor expanding the nonlinear
terms around the old values. For example, for the nonlinear term in Z, we write

f ≡ zk+1 xk+1 = f(xk, zk) + (xk+1 − xk)
(

∂f

∂x

)
xk,zk

+ (zk+1 − zk)
(

∂f

∂z

)
xk,zk

+· · ·

= xk zk + (xk+1 − xk) zk + (zk+1 − zk) xk + · · · .

Substituting this expansion into (9.11) and collecting terms yields

xk+1 + h yk+1 + h zk+1 = xk,

−hxk+1 + (1 − h a) yk+1 = yk,

−h zk xk+1 + (1 + h c − hxk) zk+1 = h b + (1 − hxk) zk.

(9.12)

This system of three linear algebraic equations is then solved on each time step.
The semi-implicit scheme derived above is only of O(h) accuracy. To derive

a second-order accurate implicit scheme, we can average3 the old and the new
3This is the basis of the Crank–Nicolson scheme for solving nonlinear PDEs.

on the rhs, i.e., replace equations (9.11) with

xk+1 =xk+
h

2
(Xk+Xk+1), yk+1 =yk+

h

2
(Yk+Yk+1), zk+1 =zk+

h

2
(Zk+Zk+1).

(9.13)
Using the same expansion for the nonlinear term, a second-order semi-implicit
algorithm for solving the Rössler equations then is

xk+1 +
1
2
h yk+1 +

1
2
h zk+1 = xk − 1

2
h (yk + zk),

−1
2
hxk+1 + (1 − 1

2
h a) yk+1 =

1
2
hxk + (1 +

1
2
h a) yk,

−1
2
h zk xk+1 + (1 +

1
2
h c − 1

2
hxk) zk+1 = h b + (1 − 1

2
h c) zk.

(9.14)

In this recipe, we will iterate equations (9.14), taking 10 digits,
> restart: with(plots): Digits:=10:

with a=b=0.2, c=5.0, h=0.05 and n=3000 iterations.
> a:=0.2: b:=0.2: c:=5.0: h:=0.05: n:=3000:

The initial condition at t0 =0 is x0 =−1, y0 =0, z0 =0.
> t[0]:=0: x[0]:=-1: y[0]:=0: z[0]:=0:

After recording the start time, a do loop is created to iterate eqs. (9.14)
> begin:=time():

> for k from 0 to n do

The rhs of each equation is entered in A1[k], A2[k], and A3[k].
> A1[k]:=x[k]-0.5*h*(y[k]+z[k]);

> A2[k]:=0.5*h*x[k]+(1+0.5*h*a)*y[k];

> A3[k]:=h*b+(1-0.5*h*c)*z[k];

The three equations of (9.14) are entered in E1, E2, and E3.
> E1:=x[k+1]+0.5*h*y[k+1]+0.5*h*z[k+1]=A1[k];

> E2:=-0.5*h*x[k+1]+(1-0.5*h*a)*y[k+1]=A2[k];

> E3:=-0.5*h*z[k]*x[k+1]+(1+0.5*h*c-0.5*h*x[k])*z[k+1]=A3[k];

The three algebraic equations, E1, E2, E3 are numerically solved for xk+1, yk+1,
and zk+1 and the solution then assigned.

> sol[k+1]:=(fsolve({E1,E2,E3},{x[k+1],y[k+1],z[k+1]}));
> assign(sol[k+1]);

The values of xk+1, yk+1, zk+1, and tk+1 = tk + h, are recorded,
> x[k+1],y[k+1],z[k+1]; t[k+1]:=t[k]+h;

and the plotting point (xk, yk, zk) formed for the kth time step.
> pt[k]:=[x[k],y[k],z[k]];

> end do:

On ending the do loop, the cpu time for the loop is calculated
> cpu:=time()-begin;

cpu := 10.242
and found to be about 10 seconds. Using the spacecurve command with a line
style and zhue coloring, the solution is plotted and shown in Figure 9.6.

> spacecurve([seq(pt[j],j=0..n)],style=line,shading=zhue,

orientation=[45,60],axes=boxed,labels=["x","y","z"]);

Figure 9.6: Rössler’s strange attractor.

After an initial transient period, the trajectory is attracted to a localized region,
where it continually traces out a new path, i.e., the motion is chaotic. This
localized chaotic motion, resembling a man’s “top hat”, is an example of what
mathematicians call a strange attractor. Starting with some other initial values
of x0, y0, and z0, you will find that the trajectory is eventually attracted to
the top hat region. For other values of the parameters, periodic motions can
also be observed. As an exercise, try varying the parameter c, holding all other
parameter values fixed.

9.2 Partial Differential Equations

Finite difference approximations can also be used to numerically solve PDEs.
For example, our first recipe involves finding the steady-state temperature
T (x, y) inside a thin rectangular plate with specified temperatures on the four
edges. T (x, y) will satisfy the 2-dimensional form of Laplace’s equation,

∂2T

∂x2 +
∂2T

∂y2 = 0. (9.15)

Our approach is to divide the x-y plane into a rectangular grid or “mesh”, as
shown in Figure 9.7, with each small rectangle having sides of length h and k.

Figure 9.7: Subdividing the x-y plane with a numerical mesh.

The coordinates of a typical mesh (intersection) point P are x = i h, y = j k,
with i, j = 0, 1, 2, .... The mesh points may be labeled by a pair of integers,
the point P being indicated by (i, j), and the temperature at that point by
Ti,j ≡ T (x= i h, y = j k). From Chapter 2, each second derivative in Laplace’s
equation can be replaced with a central difference approximation (CDA), viz.,

(Ti+1,j − 2 Ti,j + Ti−1,j)
h2 +

(Ti,j+1 − 2 Ti,j + Ti,j−1)
k2 = 0, (9.16)

or, on setting r=(h/k)2 and rearranging,

2 (1 + r) Ti,j − Ti+1,j − Ti−1,j − r Ti,j+1 − r Ti,j−1 = 0. (9.17)

For a given point P , the mesh points involved in (9.17) are as shown in Fig-
ure 9.7. If the temperature is specified at the mesh points making up the
boundary of the plate, equation (9.17) can be used to calculate the tempera-
ture at each of the internal mesh points, as will now be demonstrated.

9.2.1 Steady-State Temperature Distribution

If you can’t stand the heat, get out of the kitchen.
Harry S. Truman, former American president, (1884–1972)

A thin rectangular plate, with 0 ≤ x ≤ L1 = 1/2 and 0 ≤ y ≤ L2 = 1, has the
following temperature distributions along its four edges: T (x, 0)=500 x (L1−x),
T (x, L2)=700 x (L1 − x), T (0, y)=0, T (L1, y)=1000 y (L2 − y). Dividing the
plate into a numerical mesh with 15 steps in the x direction and 30 steps in the
y direction, determine and plot the temperature profile inside the plate.

The plots library package is loaded and the start time recorded.
> restart: with(plots): begin:=time():

The values L1=1/2, L2=1, and the numbers of steps, m=15 and n=30, are
entered. The corresponding stepsizes h = L1/m and k = L2/n are calculated,
along with the ratio r=(h/k)2.

> L1:=1/2: L2:=1: m:=15: n:=30: h:=L1/m; k:=L2/n; r:=(h/k)ˆ2;

h :=
1
30

k :=
1
30

r := 1

The x and y coordinates of the mesh points are generated.
> Xcoords:=seq(x[i]=i*h,i=0..m): Ycoords:=seq(y[j]=j*k,j=0..n):

The temperature distributions along the four edges are evaluated at the bound-
ary mesh points by setting x= i h and y=j k.

> bc1:=seq(T[i,0]=500*i*h*(L1-i*h),i=0..m):

> bc2:=seq(T[i,n]=700*i*h*(L1-i*h),i=0..m):

> bc3:=seq(T[0,j]=0,j=0..n):

> bc4:=seq(T[m,j]=1000*j*k*(L2-j*k),j=0..n):

The x and y mesh point coordinates and the 4 boundary conditions are assigned.
> assign(Xcoords,Ycoords,bc1,bc2,bc3,bc4):

A functional operator f is formed to calculate the lhs of equation (9.17).
> f:=(i,j)->2*(1+r)*T[i,j]-T[i+1,j]-T[i-1,j]-r*T[i,j+1]

-r*T[i,j-1];

f := (i, j) → 2 (1 + r) Ti, j − Ti+1, j − Ti−1, j − r Ti, j+1 − r Ti, j−1

Making use of f and a nested sequence command, the equations which have to
be solved at the (m − 1) × (n − 1) internal mesh points are generated.

> eqs:={seq(seq(f(i,j)=0,i=1..m-1),j=1..n-1)}:
The (m − 1) × (n − 1) unknown temperature variables Ti,j are entered.

> vars:={seq(seq(T[i,j],i=1..m-1),j=1..n-1)}:
The mesh equations are then numerically solved for the variables, the answers
being given to 6 digits. The solution is then assigned.

> sol:=evalf(fsolve(eqs,vars),6): assign(sol):

We now create the plotting points (xi, yj , Ti,j) for i=0 to m and j =0 to n.
> pts:=seq(seq([x[i],y[j],T[i,j]],i=0..m),j=0..n):

The numerical points are plotted with the pointplot3d command, being rep-
resented by size 6 circles which are colored with the zhue shading option. The
resulting temperature profile inside the 3-dimensional viewing box is as shown
in Figure 9.8. The box can be rotated in the usual manner.

> pointplot3d([pts],symbol=circle,symbolsize=6,shading=zhue,

axes=boxed,orientation=[55,60],labels=["x","y","T"]);

Figure 9.8: Numerically obtained temperature profile inside plate.

The cpu time for the entire recipe is about 13 seconds.
> cpu:=time()-begin;

cpu := 13.361

9.2.2 1-Dimensional Heat Flow

In general, the art of government consists in taking as much money
as possible from one party of the citizens to give to the other.
Francois Voltaire, French philosopher, (1696–1778)

One-dimensional heat flow is governed by the temperature diffusion equation

∂T

∂t
= σ

∂2T

∂x2 , (9.18)

with x the spatial coordinate, t the time, and σ the heat diffusion coefficient.
Setting y=σ t, a numerical algorithm can be formed by replacing the second

order x derivative with a CDA (spatial step, h) and the first order y derivative
by a FWDA (y step, k), viz.,

Ti,j+1 − Ti,j

k
=

Ti+1,j − 2 Ti,j + Ti−1,j

h2 , (9.19)

or, on setting r = k/h2 and rearranging,

Ti,j+1 = r Ti−1,j + (1 − 2 r) Ti,j + r Ti+1,j . (9.20)

In terms of a “mesh diagram” for the x-y plane, the mesh points involved in
(9.20) are as shown in Figure 9.9. The unknown temperature Ti,j+1 is to be
explicitly determined on the j + 1st time step from the three known values
Ti−1,j , Ti,j , and Ti+1,j on the previous jth step.

y = j k

x = i h
T (x,0) specified

T( 0,y) = 0

T i, 0

T           T            Ti−1, j         i, j             i+1, j

T i, j+ 1

L

T (L, y ) = 0

0

Figure 9.9: Numerical mesh for the heat flow recipe.

One starts with the bottom row (j =0) which corresponds to the specified initial
temperature distribution inside, say, a thin rod of length L, and calculates T
at each internal mesh point of the first (j =1) time row. With Ti,1 all known,
one proceeds to calculate T on the second time row, and so on. To implement
the algorithm, we must also specify the boundary conditions at the ends of the
rod. For example, we might have T (0, y)=T (L, y)= 0 as in the figure.

Suppose that T (x, 0) = 200x3 (L − x) with L = 1 m. To implement the
algorithm, the LinearAlgebra package is loaded so a matrix multiplication can
be used for calculating T at the internal mesh points. Let’s subdivide the range
x=0 to x=L by taking N =25 internal grid points. The relevant matrix A for
implementing the rhs of (9.20) will be an N ×N tridiagonal matrix, with 1−2 r
along the main diagonal, r along the first subdiagonals, and zeros everywhere
else. For N > 10, the matrix will not be explicitly displayed unless rtablesize
is set to N or larger in the interface command. Here it is set to infinity.

> restart: with(LinearAlgebra): interface(rtablesize=infinity):

Entering N and L, the spatial stepsize is h=L/(N +1) which is now evaluated.
The explicit scheme becomes numerically unstable if r = k/h2 > 1/2, so let’s
take r =0.05. Then the y stepsize k = r h2 is calculated. The number of plots
to be produced in the iteration is given by numplots.

> N:=25: L:=1.0: h:=L/(N+1); r:=0.05: k:=r*hˆ2; numplots:=100:

h := 0.03846153846 k := 0.00007396449705
Using the BandMatrix command, the relevant matrix A is entered. The diagonal
entries are given by 1 − 2 r = 0.90, the subdiagonal entries by r = 0.05. The
number 1 indicates that there is one subdiagonal and N gives the size of the
matrix, i.e., N × N . For brevity, the output shown here in the text has been
truncated, only 5 of the 25 rows being shown.

> A:=BandMatrix([r,1-2*r,r],1,N);

A:= [0.90, 0.05, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0]
[0.05, 0.90, 0.05, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0]
[0, 0.05, 0.90, 0.05, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0]
[...............................................................................................................]
[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0.05, 0.90, 0.05]
[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0.05, 0.90]

A functional operator f is formed for generating the values of the initial tem-
perature profile at the internal spatial grid points.

> f:=x->evalf(200*xˆ3*(L-x)):

Using f, the values of the temperature at the internal x mesh points on the
zeroth time row are formed into a column vector v. Again, only a few of the 25
entries are shown here in the text.

> v:=Vector([[seq(f(i*h),i=1..N)]]);

v :=

⎡
⎢⎢⎢⎢⎢⎢⎢⎢⎢⎢⎢⎢⎢⎢⎣

0.01094149364
0.08403067118
0.2717867022
..................
..................
21.01335738
..................
..................
12.10041666
6.838433534

⎤
⎥⎥⎥⎥⎥⎥⎥⎥⎥⎥⎥⎥⎥⎥⎦

To establish the vertical scale of our plot, the following max command is used
to select the largest temperature value from the sequence of entries in v.

> vmax:=max(seq(v[i],i=1..N));

vmax := 21.01335738
The start time for the following do loop, which carries out the requisite matrix
multiplication, is recorded.

> begin:=time():

The do loop runs from n=1 to numplots. The matrix multiplication is carried
out in T, a plot being produced on every 20th time step. For plotting purposes

the end points are included in the list of plotting points. The points will be
joined by straight lines.

> for n from 1 to numplots do

> T:=Aˆ(20*n) . v;

> p[n]:=plot([[0,0],seq([i/(N+1),T[i]],i=1..N),[1,0]],

tickmarks=[2,2],labels=["x","y"]);
> end do:

The cpu time for the do loop is about 20 s.
> cpu:=time()-begin;

cpu := 20.037
Making use of the display command with the insequence=true option, the
heat diffusion process in the rod is animated. The opening frame of the anima-
tion is shown in Figure 9.10. Execute the recipe to see what happens.

> plots[display]([seq(p[i],i=1..numplots)],insequence=true,

view=[0..L,0..vmax+1]);

Figure 9.10: Opening frame of the heat diffusion animation.

9.2.3 Von Neumann Stability Analysis

No civilization would ever have been possible
without a framework of stability....
Hannah Arendt, American political philosopher, (1906–75)

That the explicit scheme used in the heat diffusion equation recipe 09-2-2 is
conditionally stable, i.e., stable for r=k/h2 ≤ 1/2, could be proved by executing
the recipe with varying values of r and looking at the numerical output. Alter-
nately, one can use the Von Neumann stability analysis as will be demonstrated
in the following recipe.

Recall that the relevant algorithm is of the structure

Tm,n+1 = r Tm−1,n + (1 − 2 r) Tm,n + r Tm+1,n (9.21)

with m and n indexing the spatial and time steps, respectively. Assume that
Tm,n = T 0

m,n + Um,n, where T 0 is the exact solution of the difference scheme
and U represents a small numerical error due to roundoff, etc. Substituting this
form into (9.21) yields an identical finite difference scheme for U . The question
is “Does U grow or decay with time?”, being unstable (stable) if it grows (de-
cays). To answer this question, we assume that Um,n can be represented by the
following exponential form,

Um,n = ei m θ ei n λ, (9.22)

with θ real and λ= α + i β. It is assumed that both α and β are real.
> restart: assume(alpha::real,beta::real):

An operator U is formed to calculate Um,n for arbitrary subscripts m and n.
> U:=(m,n)->exp(I*m*theta)*exp(I*n*lambda);

U := (m, n) → e(m θ I) e(n λ I)

Eq. (9.21) is entered in terms of U , each term being automatically evaluated.
> eq:=U(m,n+1)=r*U(m-1,n)+(1-2*r)*U(m,n)+r*U(m+1,n);

eq := e(m θ I) e((n+1) λ I) = r e((m−1) θ I) e(n λ I)

+(1 − 2 r) e(m θ I) e(n λ I) + r e((m+1) θ I) e(n λ I)

Then, eq is divided by Um,n and simplified.
> eq2:=simplify(eq/U(m,n));

eq2 := e(λ I) = 2 r cos(θ) + 1 − 2 r

We substitute λ=α + i β into eq2 .
> eq3:=subs(lambda=alpha+I*beta,eq2);

eq3 := e((α+β I) I) = 2 r cos(θ) + 1 − 2 r

The absolute values of the lhs and rhs of eq3 are equated, and simplified.
> eq4:=simplify(abs(lhs(eq3)))=abs(rhs(eq3));

eq4 := e(−β) = |2 r cos(θ) + 1 − 2 r|
For stability, β ≥ 0 is required, because if β < 0, then Um,n ∼ e|β| n diverges as
n (i.e., time) increases. For β ≥ 0, e−β ≤ 1, so we must have the rhs of eq4 less
than or equal to 1 for stability. The rhs of eq4 is turned into an operator g in
terms of r by using the unapply command.

> g:=unapply(rhs(eq4),r);

g := r → |2 r cos(θ) + 1 − 2 r|
Using g, the rhs of eq4 is plotted over the range θ=−2.5 π to 2.5π for r=0.49
and 0.51, being represented on the computer screen by red and blue curves,
respectively. A horizontal green line is also plotted at a vertical height 1. The
corresponding black and white picture is shown in Figure 9.11.

> plot([g(.49),g(.51),1],theta=-2.5*Pi..2.5*Pi,

color=[red,blue,green],numpoints=1000);

0

1

–6 theta 6

Figure 9.11: The r=0.51 curve > 1 for some θ, the r=0.49 curve doesn’t.

Since the r=0.51 curve exceeds 1 for some ranges of θ, the numerical scheme is
unstable for this r value. The r=0.49 curve doesn’t exceed 1, so the algorithm
is stable. By varying r between 0.49 and 0.51, you may use the recipe to check
that the critical value of r for numerical stability or instability is 1/2.

9.2.4 Sometimes It Pays to be Backwards

Diaper backward spells repaid. Think about it.
M. McLuhan, Canadian communications theorist, Vancouver Sun, June 1969

An unconditionally stable algorithm can be obtained for ∂T/∂y = ∂2T/∂x2 by
approximating the y derivative with the BWDA. In this case, on still replacing
the x derivative with the standard CDA, the numerical algorithm becomes

(1 + 2 r) Ti,j − r Ti+1,j − r Ti−1,j = Ti,j−1. (9.23)
Let’s use equation (9.23) to resolve the example given in recipe 09-2-2, which
involved determining the evolution of the temperature distribution in a thin
rod of length L=1 m, with boundary conditions T (0, y)=T (L, y)= 0, and an
initial profile T (x, 0)=f(x)=200 x3 (L − x).

We take M = 30 spatial steps, so h = L/M = 1/30. Choosing the time (y)
step to be k =0.002 produces a ratio r =k/h2 =1.8. Not only is r larger than
1/2, it’s greater than 1! Nevertheless, as you will see, the scheme remains stable,
although as r is increased the algorithm becomes increasingly less accurate. The
total number of time steps is taken to be N =250.

> restart: with(plots):

> L:=1: M:=30: h:=L/M; k:=0.002; r:=k/hˆ2; N:=250:

h :=
1
30

k := 0.002 r := 1.800

The values of xi = i h are calculated in Xcoords for i=0 to M , and the sequence
of numbers assigned.

> Xcoords:=seq(x[i]=i*h,i=0..M): assign(Xcoords):

A functional operator f is introduced for generating the initial temperature
value at the ith spatial mesh point.

> f:=i->evalf(200*(x[i]ˆ3*(L-x[i]))):

Using f, the sequence of initial T values at the grid points is calculated.

> ic:=seq(T[i,0]=f(i),i=0..M):

The two boundary conditions are entered,

> bc1:=seq(T[0,j]=0,j=0..N): bc2:=seq(T[M,j]=0,j=0..N):

and are assigned, along with the initial condition.

> assign(ic,bc1,bc2):

An operator F is formed for implementing the algorithm (9.23).

> F:=(i,j)->(1+2*r)*T[i,j]-r*T[i+1,j]-r*T[i-1,j]-T[i,j-1];

F := (i, j) → (1 + 2 r) Ti, j − r Ti+1, j − r Ti−1, j − Ti, j−1

The start time for executing the algorithm in the do loop is recorded.

> begin:=time():

We now generate and numerically solve the algorithm equations, F (i, j)=0, at
each of the internal grid points on each of the j time steps, for j =1 to N .

> for j from 1 to N do

> eqs[j]:={seq(F(i,j)=0,i=1..M-1)};
> sol[j]:=fsolve(eqs[j],{seq(T[i,j],i=1..M-1)});
> assign(sol[j]):

> end do:

The cpu time for the do loop is now displayed.

> cpu:=time()-begin;

cpu := 14.110
A graphing operator gr is created for producing the sequence (i = 0 to M) of
plotting points (xi, Ti,j) on the jth time step and plotting them.

> gr:=j->plot([seq([x[i],T[i,j]],i=0..M)]):

The temperature diffusion process is animated with the display command,
the insequence=true option being included. Click on the computer plot and
the start arrow in the tool bar to see the animation.

> display([seq(gr(j),j=0..N)],insequence=true,labels=["x","T"]);

9.2.5 Daniel Still Separates, I Now Iterate

The ultimate result of shielding men from the effects of folly
is to fill the world with fools.
Herbert Spencer, British philosopher, sociologist, (1820–1903)

To numerically solve the 1-dimensional wave equation, we can replace each
second derivative with a CDA, viz.,

∂2U

∂x2 =
1
c2

∂2U

∂t2
=⇒ Ui+1,j − 2 Ui,j + Ui−1,j

h2 =
1
c2

Ui,j+1 − 2 Ui,j + Ui,j−1

k2 ,

=⇒ Ui,j+1 = 2 (1− r) Ui,j + r (Ui+1,j +Ui−1,j)−Ui,j−1, r=(c k/h)2. (9.24)

In this case, the unknown Ui,j+1 depends explicitly on U values on the previous
two steps, as shown schematically in Figure 9.12, where x is horizontal and
t vertical. To start the iteration scheme (9.24), we must therefore know the

U U Ui−1,j         i,j              i+1,j

Ui, j+1

Ui, j −1

Figure 9.12: Mesh points involved in wave equation algorithm (9.24).

values of U on both the j = 0 and j = 1 time steps. To see how we deal with
this issue, let’s consider the following possible scenario.

Young Daniel is playing with a light horizontal string of length L fixed at
x=0 and L, i.e., U(0, t)=U(L, t)=0. Before he manages to separate it into its
individual strands, his older cousin Justine cleverly manages to give it an initial
profile U(x, 0) = f(x) = A x3 (L − x) and transverse velocity U̇(x, 0) = g(x) =
B x (L−x)2. Our task is to iterate the numerical algorithm (9.24) to determine
the subsequent motion of the string and then animate the string vibrations,
taking L=1 m, A=10 m−3, B=5 m−2, and c=3 m/s.

On the zeroth time row, we have Ui,0 =f(xi). Using the forward difference
approximation, the U values on the first time row, i.e., Ui,1, are found thus:

∂U(xi, 0)
∂t

=
Ui,1 − Ui,0

k
= g(xi), so Ui,1 = f(xi) + k g(xi) + O(k2). (9.25)

An even better approximation for Ui,1, which we shall use in our recipe, is

Ui,1 = f(xi) + k g(xi) +
r

2
[f(xi+1) − 2 f(xi) + f(xi−1)] + O(k3). (9.26)

To begin the recipe, the values of L, c, A, and B are entered. The total time is
taken to be T =2 s, and the number of time and spatial steps are N =200 and
M =20, respectively.

> restart: with(plots):

> L:=1: T:=2: M:=20: N:=200: c:=3: A:=10: B:=5:

The spatial step h=L/M , time step k=T/N , and r=(c k/h)2 are calculated.
> h:=evalf(L/M); k:=evalf(T/N); r:=evalf((c*k/h)ˆ2);

h := 0.05000000000 k := 0.01000000000 r := 0.3600000000
Here, we have r = 0.36. Using the Von Neumann stability analysis, it can be
shown that the condition for stability is that r ≤ 1, referred to as the Courant
stability condition.

The boundary conditions U0,j = 0 and UM,j = 0 for j = 0 to N are entered
in bc1 and bc2, which are both assigned.

> bc1:=seq(U[0,j]=0,j=0..N): assign(bc1):

> bc2:=seq(U[M,j]=0,j=0..N): assign(bc2):

Functional operators f and g are introduced for calculating f(x) and g(x) at
the internal grid points xi = i h.

> f:=x->evalf(A*xˆ3*(L-x)): g:=x->evalf(B*x*(L-x)ˆ2):

The sequence of values Ui,0 =f(i h) are generated in ic1 for i=1...M − 1 and
the result assigned. Similarly, the sequence of values Ui,1 are generated using
(9.26) in ic2, and the result also assigned.

> ic1:=seq(U[i,0]=f(i*h),i=1..M-1): assign(ic1):

> ic2:=seq(U[i,1]=f(i*h)+k*g(i*h)

+(r/2)*(f((i+1)*h)-2*f(i*h)+f((i-1)*h)),i=1..M-1):
> assign(ic2):

The time at the beginning of the do loop is recorded.
> begin:=time():

The algorithm (9.24) is iterated from j =1 to N − 1 and the solution assigned.
> for j from 1 to N-1 do

> sol:=seq(U[i,j+1]=2*(1-r)*U[i,j]+r*(U[i+1,j]+U[i-1,j])

-U[i,j-1],i=1..M-1);
> assign(sol):

> end do:

The cpu time for the do loop is found to be about 0.4 s.
> cpu:=time()-begin;

cpu := 0.404
An operator gr is formed for plotting the spatial profile on the jth time step.

> gr:=j->plot([seq([i*h,U[i,j]],i=0..M)],labels=["x","U"]):

The motion is animated by including insequence=true in the display com-
mand. The opening frame of the animation is shown in Figure 9.13.

> display(seq(gr(j),j=0..N),insequence=true);

–1

–0.5

0

0.5

1

U

0.2 0.4 0.6 0.8 1
x

Figure 9.13: Opening frame of the vibrating string animation.

9.2.6 Interacting Laser Beams

The exercise of power is determined by thousands of interactions
between the world of the powerful and that of the powerless,
all the more so because these worlds are never divided by
a sharp line: everyone has a small part of himself in both.
Vclav Havel, Czech playwright, president, (b. 1936)

The interaction of two intense laser pulses as they pass through each other in
opposite directions in a certain resonant absorbing fluid can be described [RE76]
by the following normalized PDEs for the laser intensities u and v:

∂u

∂x
+

∂u

∂y
= −g1 u v − b u,

∂v

∂x
− ∂v

∂y
= −g2 u v + b v. (9.27)

Here x is the distance inside the fluid medium, y the time, g1 > 0 and g2 > 0
are the “gain” coefficients, and b ≥ 0 the absorption coefficient. The u (v) beam
travels in the positive (negative) x direction.

Our goal is to numerically solve this set of nonlinear PDEs for the intensities,
using the method of characteristics, which is based on finding “characteristic”
directions along which the PDEs can be reduced to ODEs. Since u=u(x, y) and
v=v(x, y), then du=(∂u/∂x) dx+(∂u/∂y) dy and dv=(∂v/∂x) dx+(∂v/∂y) dy.

Making use of these results, equations (9.27) can be rewritten as

du

dx
+ (1 − dy

dx
)
∂u

∂y
= f1,

dv

dx
− (1 +

dy

dx
)
∂v

∂y
= f2, (9.28)

with f1 ≡ −g1 u v − b u and f2 ≡ −g2 u v + b v. If we work along a line whose
slope is dy/dx = 1, then du/dx = f1, while along a line of slope dy/dx = −1
one has dv/dx=f2. The characteristic directions dy/dx=±1 form a diamond-
shaped grid as in Figure 9.14, the grid spacing ∆y=∆x=h. To solve for u(x, y)

∆x=h

L                 R

P

x

y

∆y=h
j=0

j=1

j=2

i=0 i=1 i=2

(1,1)           (3,1)           (5,1)

(2,0)            (4,0)          (6,0)

(0,2)          (2,2)           (4,2)

x=0                                                                                     x=L=1

Figure 9.14: Diamond-shaped grid for solving (9.27).

and v(x, y), we impose the following initial and boundary conditions:

• u(x, 0)=v(x, 0)=0 for 0 < x < L=1, i.e., no pulses initially inside fluid,

• u(0, y)=v(1, y)=sin(2π y) for 0 ≤ y ≤ y0 =1/2 and zero for y > y0, i.e.,
positive half-sine wave pulses fed in at x=0 and x=L=1.

Starting on the bottom time row, we move along the characteristic directions
dy/dx = 1 (e.g., from the point L to P ) and dy/dx = −1 (from R to P ) in
calculating the changes in u and v, respectively. Using, say, the Euler approxi-
mation, the values of u and v at P are given by

uP = uL + h (f1)L, vP = vR − h (f2)R. (9.29)

Taking, say, g1 =0.4, g2 =20, b=0.5, we implement (9.29) in the following
recipe and animate the numerical result. The fluid sample length L = 1 is
divided into M = 120 steps, so that the x (or y) stepsize is h = L/M = 1/120.

The leading edge of each pulse will just make contact in N = M/2 = 60 time
steps. If we let i and j index the x and y steps, as in Figure 9.14, the grid
coordinates for j =0, 2, ... are (0, j), (2, j), (4, j), ..., while for j =1, 3, ... they
are (1, j), (3, j), (5, j), ... For later convenience, we set m=2 i and n=2 j.

> restart: with(plots): begin:=time():

> L:=1: M:=120: h:=L/M; N:=M/2; m:=2*i: n:=2*j:

h :=
1

120
N := 60

The values of g1, g2, and b are entered. A functional operator S is introduced
to calculate the input laser shape on the boundaries.

> g1:=0.4; g2:=20; b:=0.5; S:=j->sin(n*evalf(Pi)/N);

g1 := 0.4 g2 := 20 b := 0.5 S := j → sin(
n evalf(π)

N
)

Using S, the boundary conditions are applied at i=0 and M in bc1 and bc2.
> bc1:=seq(u[0,n]=S(j),j=0..N/2),seq(u[0,n]=0,j=N/2+1..3*N):

> bc2:=seq(v[M,n]=S(j),j=0..N/2),seq(v[M,n]=0,j=N/2+1..3*N):

The initial condition is entered and assigned along with bc1 and bc2.
> ic:=seq(u[m,0]=0,i=0..M/2),seq(v[m,0]=0,i=0..M/2):

> assign(bc1,bc2,ic):

Functional operators are formed to evaluate f1 and f2 for a given i, j.
> f1:=(i,j)->-g1*u[i,j]*v[i,j]-b*u[i,j]:

f2:=(i,j)->-g2*u[i,j]*v[i,j]+b*v[i,j]:
A double do loop is used to iterate (9.29). The conditional if...then statement
is included so that i starts at i0=0 for j =0, 2, 4, ... and i0=1 for j =1, 3, ....

> for j from 0 to 3*N do

> if j mod 2=0 then i0:=0 else i0:=1 end if;

> for i from i0 to M by 2 do

> u[i+1,j+1]:=u[i,j]+h*f1(i,j);

> v[i-1,j+1]:=v[i,j]-h*f2(i,j);

> end do: end do:

An operator gr is formed to plot u and v on the jth time step.
> gr:=j->plot([[seq([m*h,u[m,n]],i=0..M/2)],[seq([m*h,v[m,n]],

i=0..M/2)]],color=[red,blue],labels=["x","u,v"],thickness=2):
The numerically obtained profiles are animated with the display command
with the insequence=true option. Click on the computer plot and the start
arrow to initiate the animation.

> display([seq(gr(j),j=0..2*N)],insequence=true);

The cpu time for the entire recipe is given below.
> cpu:=time()-begin;

cpu := 7.752

9.2.7 KdV Solitons

In this recipe, we shall numerically investigate the collision of a taller faster
solitary wave overtaking a shorter slower one. The standard algorithm for nu-
merically integrating the KdV equation (8.25) is obtained as follows. Returning
to the general Taylor expansion (9.3) of y(x ± h), adding the plus and minus
results yields the central difference approximation to the first derivative, viz.,

y ′ =
y(x + h) − y(x − h)

2 h
+ O(h2). (9.30)

This approximation is used for the first derivatives in (8.25), viz.,
∂U

∂t
⇒ (Ui,j+1 − Ui,j−1)

2 k
,

∂U

∂x
⇒ (Ui+1,j − Ui−1,j)

2 h
,

where k and h are the time and spatial stepsizes, respectively.
A central difference approximation to the third derivative can be obtained

by modifying the first part of Recipe 02-1-2 as follows. A functional operator
t is formed to Taylor expand y(x + a h) to 5th order in h for arbitrary a. The
order of term is removed with the convert( ,polynom) command.

> restart:

> t:=a->y(x+a*h)=convert(taylor(y(x+a*h),h,5),polynom):

Then, using the operator, the sum y(x+2 h)−2 y(x+h)+2 y(x−h)−y(x−2 h)
is calculated in eq1 , yielding 2h3 y ′′′ on the rhs. The error here is O(h5).

> eq1:=t(2)-2*t(1)+2*t(-1)-t(-2);

eq1 := y(x + 2 h) − 2 y(x + h) + 2 y(x − h) − y(x − 2 h) = 2 (D(3))(y)(x) h3

A finite difference approximation with error O(h2) follows for the third deriva-
tive on using the isolate command.

> eq2:=isolate(eq1,D[1,1,1](y)(x));

eq2 := (D(3))(y)(x) = −1
2

−y(x + 2 h) + 2 y(x + h) − 2 y(x − h) + y(x − 2 h)
h3

So, we will make the central difference approximation,
∂3U

∂x3 ⇒ (Ui+2,j − 2 Ui+1,j + 2 Ui−1,j − Ui−2,j)
2 h3 .

Finally, let’s use the central difference approximation (Ui+1,j +Ui,j +Ui−1,j)/3,
which also has an error O(h2), for the U term in the KdV equation. Putting
all these approximations together, the algorithm for the KdV equation is

Ui,j+1 = Ui,j−1 − k

h

(Ui+1,j + Ui,j + Ui−1,j)
3

(Ui+1,j − Ui−1,j)

− k

h3 (Ui+2,j − 2Ui+1,j + 2Ui−1,j − Ui−2,j). (9.31)

Using the Von Neumann stability analysis, it can be shown that this algorithm
is numerically stable for k/h3 < 2/(3

√
3)=0.3849.

Note that the algorithm connects the j +1 time step to steps j and j −1, so
to iterate (9.31) the U values must be known on the j =0 and j =1 time steps.

The former are given by Ui,0 =f(xi), where f(x) is the input profile. For the U
values on the first step, we shall use the result (9.25), i.e., Ui,1 =f(xi)+k g(xi),
where g(x) is the initial transverse velocity.

We begin the main part of our recipe by considering M =150 spatial steps
and N =300 time steps. The starting time of the recipe is also recorded.

> restart: with(plots): M:=150: N:=300: begin:=time():

A functional operator F is formed to calculate a time-dependent solitary wave
profile centered at time t=0 at x=X and having speed c.

> F:=(X,c)->3*c *(sech((sqrt(c)/2)*((x-X)-c*t)))ˆ2:

The spatial stepsize is taken to be h=1, the time stepsize k =0.25, so k/h3 =
0.25 and the scheme will be stable. For f(x), we will consider two separated
solitary waves, a taller one having speed c1 =0.8 and a shorter one with speed
c2 = 0.1. The taller, faster, solitary wave will be initially centered at X1 =
M/3=50, the shorter, slower, wave at X2 =M/2=75. As time progresses, the
taller solitary wave will overtake the shorter one and a collision take place.

> h:=1: k:=.25: c1:=.8; c2:=.1; X1:=M/3; X2:=M/2;

c1 := 0.8 c2 := 0.1 X1 := 50 X2 := 75
Then F(X1,c1)+F(X2,c2) is entered and the time derivative taken in g.

> f:= F(X1,c1)+F(X2,c2): g:=diff(f,t):

On setting t = 0, the initial profile is given by f = 3 c1 sech(
√

c1
2 (x−X1 ))2 +

3 c2 sech(
√

c2
2 (x−X2 ))2 which is now plotted in Figure 9.15.

> t:=0: plot(f,x=0..M);

Figure 9.15: Initial profile.

To evaluate f at the spatial mesh points on the zeroth time row and g at the
mesh points on the first time row, they are turned into functional operators in
terms of x by using the unapply command.

> f2:=unapply(f,x): g2:=unapply(g,x):

Using f2 and g2, Ui,0 and Ui,1 are calculated in ic1 and ic2 for i=0 to M .

> ic1:=seq(U(i,0)=evalf(f2(i)),i=0..M):

> ic2:=seq(U(i,1)=evalf(f2(i))+k*g2(i),i=0..M):

To avoid unknown U values creeping into the do loop, we will set all U values
equal to zero for i = 0 to M and j = 2 to N . These “initialized” zeros will be
overwritten as the do loop progresses.

> init:=seq(seq(U(i,j)=0,i=0..M),j=2..N):

Then ic1 , ic2 , and init are assigned.
> assign(ic1,ic2,init):

The algorithm (9.31) is then iterated from i=2 to M − 2 and j =1 to N . We
must start i at 2 and end at M − 2 to avoid unknown U values entering the
iterative calculation.

> for j from 1 to N do;

> for i from 2 to M-2 do

> U(i,j+1):=U(i,j-1)-(k/(3*h))*(U(i+1,j)+U(i,j)+U(i-1,j))

*(U(i+1,j)-U(i-1,j))-(k/hˆ 3)*(U(i+2,j)-2*U(i+1,j)
+2*U(i-1,j)-U(i-2,j));

> end do: end do:

An operator gr is created to plot Ui,j for i=2 to M − 2 on the jth time step.
> gr:=j->plot([seq([i,U(i,j)],i=2..M-2)]):

Every second graph is displayed and the sequence of pictures animated with the
insequence=true option. Click on the computer plot and on the start arrow
to initiate the animation.

> display([seq(gr(2*j),j=0..N/2)],insequence=true);

It is observed that the tall solitary wave passes through the shorter one and both
solitary waves appear to be unchanged after the collision, aside from a very small
numerical ripple which arises because of the relatively coarse grid which was
used and the fact that the initial solitary wave profiles overlapped slightly and
their tails were artificially truncated at the edges of the numerical grid. As the
tall pulse passes through the shorter one, you should notice that the amplitudes
do not add linearly, which would be expected from linear superposition. Finally,
the cpu time for the recipe is now recorded.

> cpu:=time()-begin;

cpu := 29.913

9.3 Supplementary Recipes
09-S01: White Dwarf Equation
Chandresekhar’s theory of white dwarf stars produces the NLODE [Cha39]

x y ′′(x) + 2 y ′(x) + x (y2 − C)3/2 = 0, with y(0) = 1, y ′(0) = 0.

Making use of the first principles and dial-up Euler methods with h = .01 and
18 - digits accuracy, numerically compute y(x) over the range 0 ≤ x ≤ 4 with

C = 0.1 and plot both numerical results together. Show that the numerical
value of y(x = 1) is exactly the same for both methods. Hint: Start at x = 0.01
to avoid any problem at the origin.

09-S02: Spruce Budworm Infestation
The sudden outbreak of the spruce budworm which can rapidly defoliate a for-
est and kill the trees can be described [LJH78] by the dimensionless ODE

ẋ(τ) = rx(1 − x

K
) − x2

1 + x2 .

Here x(τ) is proportional to the budworm population number at time τ and
the growth coefficient r and carrying capacity parameter K are positive con-
stants. The first term describes the growth of the budworm population with a
saturation effect included due to the finite forest available, while the last term
models the decrease in population due to bird predation. Taking K =300, use
the first-principles Euler method with h = 0.01 and x(0) = 0.5 to determine
the time evolution of the budworm population over the time interval t=0 to 50
for r=0.1, r=0.5, and r=1.0. Plot and discuss the results.

09-S03: A Math Example
Consider the first-order nonlinear ODE

dy/dx =
√

x (y − x) (y − 2)

with y(0) = 1. Show that an analytic closed form of solution is not obtain-
able with the dsolve command. Taking h = 0.01 and 10-digit accuracy, solve
for y(x) out to x = 5 using the modified Euler method and plot every 10th
numerical point.

09-S04: Hermione Hippo
Hermione Hippo swims across the muddy Mombopo river by steadily aiming
at a tree directly across the river on the water’s edge. The Mombopo is 1 km
wide and has a speed of 1.5 km/hour while Hermione’s speed is 2 km/hour.
In Cartesian coordinates, Hermione is initially at (x = 1, y = 0) while the tree
is at (0, 0). Derive Hermione’s equations of motion in Cartesian coordinates.
Using the first-principles 4th-order RK method with h = 0.001, numerically
solve these equations to determine how long it takes Hortense to reach the tree.
Determine the analytic solution y(x) for Hortense’s path across the river and
plot the analytic and numerical solutions together in the same figure.

09-S05: The Oregonator
A kinetic model ([FKN72],[FN74],[EM00]) describing the oscillations in a cer-
tain chemical system is the Oregonator system of nonlinear ODEs, viz.,

ε ẋ = x + y − q x2 − x y, ẏ = −y + 2 h z − x y, p ż = x − z,

where x, y, and z are the dimensionless concentrations of three chemical species
and ε, p, q, and h are positive parameters. Taking ε = 0.003, p = 2, q = 0.006,
h = 0.75, x(0) = 30, y(0) = 10, and z(0) = 30, numerically solve the system of
equations using the RKF45 method with the option maxfun=0. Then use the

odeplot command with numpoints=3000 to plot (a) x vs. y vs. z for t=0..50,
(b) x vs. t, y vs. t, and z vs. t in the same figure, for t=0..30. Numerically show
that a closed loop results in the x-y-z space no matter what initial conditions
are chosen. The Oregonator is an example of a 3-dimensional limit cycle.

09-S06: Lorenz’s Butterfly
The Lorenz system [Lor63] of nonlinear ODES are given by

ẋ = σ (y − x), ẏ = r x − y − x z, ż = x y − b z,

with x, y, z real and σ, r, and b real, positive constants. Derive a second-order-
accurate semi-implicit numerical scheme for the Lorenz system. Taking σ=10,
b = 8/3, r = 28, x(0) = 2, y(0) = 5, z(0) = 5, a stepsize h = 0.01, and n = 4000
iterations, iterate the algorithm. Plot x vs. y vs. z, using a zhue shading.
The resulting figure should resemble the wings of a butterfly. After a transient
interval, the trajectory remains confined to the region of the wings, but never
retraces the same path. This is another example of a strange chaotic attractor.

09-S07: A Stiff Harmonic Oscillator
Consider the following heavily overdamped forced harmonic oscillator equation,

ẍ + 50 ẋ + x = 2 sin t,

with the initial condition x(0) = 0, ẋ(0) = 10. Solve the ODE analytically and
determine the two widely different characteristic times in the solution. Create
operators to numerically solve the ODE using the 4th-order Rungé–Kutta dial-
up method and plot the numerical solution for arbitrary stepsize h. By plotting
the analytic and numerical solution together, determine the approximate h value
at which numerical instability sets in. This critical h value should be roughly
comparable to the shortest characteristic time.

09-S08: Courant Stability Condition
Prove the Courant stability condition for the linear wave equation.

09-S09: Poisson’s Equation
Consider Poisson’s equation

∂2V

∂x2 +
∂2V

∂y2 = x ey, 0 ≤ x ≤ 2, 0 ≤ y ≤ 1,

with V (0, y)=0, V (2, y)=2 ey, V (x, 0)=x, and V (x, 1) = e x, as the boundary
conditions. Using a central difference approximation for each second derivative
at a mesh point (i, j) and evaluating the rhs of Poisson’s equation at this point,
derive a finite difference scheme for solving the equation. Dividing the x-interval
into 30 steps and the y interval into 15 steps, determine V at each mesh point
and plot the numerical solution.

09-S10: Crank–Nicolson Method
The Crank–Nicolson (CN) method results on averaging the forward and back-
ward difference methods. For the normalized linear temperature equation, the

forward and backward schemes connecting the j and j + 1 time steps are,

(Ti,j+1 − Ti,j)
k

=
(Ti+1,j − 2 Ti,j + Ti−1,j)

h2 ,

(Ti,j+1 − Ti,j)
k

=
(Ti+1,j+1 − 2 Ti,j+1 + Ti−1,j+1)

h2 .

Adding the two equations, dividing by 2, and rearranging, yields,

−r Ti−1,j+1 + 2 (1 + r) Ti,j+1 − r Ti+1,j+1 = r Ti−1,j + 2 (1 − r) Ti,j + r Ti+1,j

with r =k/h2. This finite difference CN formula relates three unknown values
of T on the j + 1 time row to three known values on the j time row. Modify
recipe 09-2-2 to implement the CN formula, taking all parameters the same,
except now with r = 0.6. As you may verify the explicit scheme in 09-2-2 is
numerically unstable for this r value, but the CN scheme is stable.

09-S11: Klein–Gordon Equation
The transverse vibrations of a horizontal string embedded in a vertical elastic
membrane which exerts a Hooke’s law restoring force on the string can be de-
scribed by the Klein–Gordon equation,

∂2U

∂x2 − ∂2U

∂t2
= a U,

with a a positive constant. If such a string of length L = 1 is fixed at both
ends and has the initial shape U(x, 0) = x4 (L − x)2 and transverse velocity
U̇(x, 0) = x (L − x), numerically determine the transverse displacement of the
string for t ≥ 0 using a matrix approach and animate the motion. Take a=100,
spatial stepsize h=0.025 and time stepsize k=0.01.



Bibliography

[AS72] M. Abramowitz and I. A. Stegun. Handbook of Mathematical Func-
tions with Formulas, Graphs, and Mathematical Tables. National
Bureau of Standards, Washington, DC, 1972.

[AW00] G. B. Arfken and H. J. Weber. Mathematical Methods for Physi-
cists, 5th ed. Academic Press, New York, 2000.

[Bay99] W. E. Baylis. Electrodynamics: A Modern Geometric Approach.

infolevel command, 19
library packages, 9
list, 7, 16
list of lists, 18, 109
mouse dragging, 31
prime notation, 22
protected symbols, 33
recipes on CD, 2

MapToBasis, 96
MatrixInverse, 121
Norm, 114
Omega, 28
Order, 70
Phi, 320
Pi, 9, 75
Re, 186, 197
Roots, 202
SetCoordinates, 101, 102
Sum, 74, 81
Transpose, 110, 118

beta, 293

filled=true, 91, 92

inifcns, 38

odeadvisor, 15, 37, 40
odeplot, 53, 311, 320

pdsolve, 129, 144, 329
phaseportrait, 302, 307
phi, 83
piecewise, 50, 74

plot, 5, 20

shading=z, 91

coupled system, 17, 25
diagnostic tool, 15
diffusion, 225
first integral, 14
first-order, 14
fourth-order, 21
Gegenbauer, 40
general solution, 14, 49, 57
Hermite, 45
homogeneous equation, 16
inhomogeneous S-L, 48
initial value problem, 16
integrating factor, 14
Legendre, 39, 65
linear, 14
Lorenz, 372
Maple advisor, 37
non-trivial solution, 22
nonlinear, 273, 292
nonlinear diffusion, 324
Oregonator, 372
prime notation, 15
Rössler, 352
Riccati, 333, 334
Schrödinger, 44

separable, 20
spruce budworm, 371
steady-state solution, 29
stiff, 349
Sturm–Liouville, 36, 281
transient solution, 29
trivial solution, 22
Van der Pol, 302, 339
variable coefficients, 36
variation of parameters, 16
Volterra–Lotka, 341
white dwarf, 370, 371
Whittaker, 45

Ogilvie, J.P., 331
Ohm’s law, 141
order of term, 39
ordinary point, 65, 306
Oregonator ODEs, 372
orthogonal contours, 188
orthogonality, 41, 72, 151, 168, 170
oscillators, driven coupled, 56
overdamping, 56

paraboloidal coordinates, 100
scale factors, 101
vector operators, 101

parameters, adjustable, 284
parametric excitation, 346
parametric solution, 258
Parseval relation, 222
Parseval’s theorem, 224, 255
partial fractions, 235
PDE

3-d diffusion, 239
diffusion, 127, 169
general solution, 179
Helmholtz, 158
Klein–Gordon, 373
Laplace, 127, 146, 151, 172, 355
linear, 127
modified diffusion, 242
nerve fiber, 331
nonlinear, 322, 328
nonlinear diffusion, 324
Poisson, 127, 156, 372

scalar Poisson, 174
Schrödinger, 127
sine-Gordon, 328
telegraph, 141
vector Poisson, 174
wave, 127

pendulum, 334
chaotic, 273
growing, 58

percolation, 324
permeability, 174
permittivity, 156
perturbation method

Lindstedt, 313
Poisson, 310

phase angle, 28
phase-plane analysis, 306
phase-plane portrait, 302, 307, 331
photorefractive material, 331
Physics of Sports, 18
piecewise function, 50
piecewise string, 180
Planck’s constant, 44
plane wave, 135
Poincaré’s theorem, 306
Poisson equation, 127, 156, 182, 334
polar, 163
pole

nth-order, 191
first-order, 191
second-order, 206
simple, 191, 203, 233
third-order, 245

polylog, 82
polynomial approximation, 59
potential, 104
potential energy, 44

minimizing, 270
power series, 65
prime notation, 15
principal argument, 185
principal branch, 209
probability amplitude, 46
probability density, 46
proton, 277

quantum oscillator, 44
quantum probability, 44

Rössler ODE, 352
radiative heat transfer, 324
radioactive chain, 55
radioactive decay, 242
radioactive gas, 242
radius of convergence, 59, 65
radius of gyration, 153
rate equation, 344
ratio test, 59
Rayleigh–Ritz method, 280, 284
recipe index file, 2
recurrence relation, 70
reflection coefficient, 135, 179
region of convergence, 206
regular, 65
regular function, 191
relaxation oscillation, 303
removable singularity, 219
residue, 191, 194, 196, 233
resistance, 141
resonance curve, 28
resonant frequency, 28
Riccati ODE, 334
right-hand rule, 87
Ritz method, 319, 335
RLC circuit, 56
roots, 218
rotating charged sphere, 178
Russell, John Scott, 322

saddle point, 306, 332
scalar field, 127
scalar Helmholtz equation, 184
scale factor, 45, 158

spherical polar, 98
Schrödinger equation, 44, 127
Schwarz–Christoffel transformation,

216
second harmonic, 180
secular term, 312, 314
separable ODE, 20
separating variables, 130

separation constant, 131, 153
separatrix, 332, 335
series

complex, 86
Fourier, 72, 74
Fourier cosine, 72, 76, 86
Fourier sine, 72, 76, 85
infinite, 83
Legendre, 72, 78
summing, 81

set, 16
shear force, 21, 42
sifting property, 33
similarity method, 327
similarity solution, 324
simple pole, 233
sine-Gordon PDE, 328
single-valued, 65, 185
singular point, 65, 306

regular, 65, 69
singularity, 57

essential, 205
removable, 219

slope, 93
slope discontinuity, 49
Snell’s law, 286
soft spring, 334
solitary wave, 322
soliton, 323

anti-kink, 329
bright, 336
dark, 336
kink, 329
Korteweg–deVries, 368
nonlocal, 331

space charge, 334
special functions, 36
specific heat, 169, 225
spherical Bessel function, 170
spherical polar operators, 99
Spiegel,Murray, 83
spring,hard, 297
spring,soft, 334
spruce budworm, 371
spruce budworm ODE, 371

squid and herring, 309
stability, 42
stability analysis

Von Neumann, 360
stable focal point, 332
stable isotope, 55
stagnation points, 219
Stark effect, 183
stationary point

focal, 306
higher-order, 306
nodal, 306
saddle, 306
simple, 306
stable, 306
unstable, 306
vortex, 306

steady state, 28, 56
stiff ODE, 349
stiff oscillator, 372
Stokes’s theorem, 124
strange attractor, 352
stream function, 188
streamlines, 188
Sturm–Liouville ODE, 36
subscript notation, 129
subsidiary conditions, 267
summand, 69
surface integral, 107
surface of revolution, 287

tangent field, 302, 308
tapered string, 180
Taylor series, 39, 206

multiple variables, 59
multivariate, 63
single variable, 59

telegraph equation, 141, 182
temperature distribution, 146

annular, 183
temperature gradient, 122
temperature switch, 181
tension, 129, 138
terminal velocity, 18, 55
thermal conductivity, 169, 225

time of descent, 260
topographical map, 92
trampoline, 149, 182
transcendental equation, 23, 24, 154
transform

Fourier, 222, 230
Fourier cosine, 222, 228
Fourier sine, 222, 225
Hankel, 253
inverse Fourier, 222, 231
inverse Laplace, 235
Laplace, 235

transient, 29
transient envelope, 56
translation, 26
transmission coefficient, 135, 179
transmission line, 141
transverse oscillation, 138
transverse velocity, 149
trial function, 280, 319
tridiagonal matrix, 357
turbulence, 18
turbulent flow, 20
turning points, 278

uncertainty principle, 255
underdamping, 28

Van Allen belts, 275
Van der Pol ODE, 302, 339

transient growth, 335
variable separation, 130, 149

incomplete, 161
variable transformation, 322, 328
variation of parameters, 16
VariationalCalculus package, 257
vector

Cartesian unit, 88
acceleration, 88
basis, 92
displacement, 88
dot product, 87
field, 87, 96
identities, 94
line integral, 104

Maple short-hand, 88
orthonormal, 124
overbar notation, 92
position, 88
scale factors, 98
sum, 87
unit, 107
vector product, 87
velocity, 88

vector field, 102, 106, 127
arrows, 105
conservative, 104

vector potential, 174, 275
VectorCalculus package, 88
velocity, 96
velocity potential, 183, 188
vibrating bar, 153
vibrating beam, 181, 256
vibrating string

energy, 180
equation of motion, 128
piecewise linear density, 135
slope continuity, 136
struck, 238
tapered, 180
whirling, 180

viewing box coordinates, 31
Volterra–Lotka ODEs, 341
vortex, 306

wave equation, 127, 129, 238
2-dimensional, 149
Fourier series, 133
general solution, 129
separating variables, 130

weight function, 168
whirling bar, 56
whirling string, 180
white dwarf stars, 370
Whittaker function, 45
Whittaker ODE, 45
wire, 42

Young’s modulus, 21, 153, 232

